

Rules for integrands of the form $(dx)^m (a + bx^2 + cx^4)^p$

x. $\int (dx)^m (bx^2 + cx^4)^p dx$

1: $\int (dx)^m (bx^2 + cx^4)^p dx$ when $p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $p \in \mathbb{Z}$, then $(bx^2 + cx^4)^p = \frac{1}{d^{2p}} (dx)^{2p} (b + cx^2)^p$

Rule 1.2.2.2.0.1: If $p \in \mathbb{Z}$, then

$$\int (dx)^m (bx^2 + cx^4)^p dx \rightarrow \frac{1}{d^{2p}} \int (dx)^{m+2p} (b + cx^2)^p dx$$

Program code:

```
(* Int[(d.*x_)^m.*(b.*x_^2+c.*x_^4)^p_,x_Symbol] :=
  1/d^(2*p)*Int[(d*x)^(m+2*p)*(b+c*x^2)^p,x] /;
FreeQ[{b,c,d,m},x] && IntegerQ[p] *)
```

2: $\int (dx)^m (bx^2 + cx^4)^p dx$ when $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(bx^2+cx^4)^p}{(dx)^{2p} (b+cx^2)^p} = 0$

Rule 1.2.2.2.0.2: If $p \notin \mathbb{Z}$, then

$$\int (dx)^m (bx^2 + cx^4)^p dx \rightarrow \frac{(bx^2 + cx^4)^p}{(dx)^{2p} (b + cx^2)^p} \int (dx)^{m+2p} (b + cx^2)^p dx$$

Program code:

```
(* Int[(d_.**x_)^m_.*(b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  (b**x^2+c**x^4)^p/((d**x)^(2*p)*(b+c*x^2)^p)*Int[(d**x)^(m+2*p)*(b+c*x^2)^p,x] /;
FreeQ[{b,c,d,m,p},x] && Not[IntegerQ[p]] *)
```

1: $\int x (a + b x^2 + c x^4)^p dx$

Derivation: Integration by substitution

Basis: $x F[x^2] = \frac{1}{2} \text{Subst}[F[x], x, x^2] \partial_x x^2$

Rule 1.2.2.2.1:

$$\int x (a + b x^2 + c x^4)^p dx \rightarrow \frac{1}{2} \text{Subst}\left[\int (a + b x + c x^2)^p dx, x, x^2\right]$$

Program code:

```
Int[x*(a+b.*x^2+c.*x^4)^p_,x_Symbol] :=
  1/2*Subst[Int[(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,p},x]
```

2: $\int (dx)^m (a + b x^2 + c x^4)^p dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.2.2.2: If $p \in \mathbb{Z}^+$, then

$$\int (dx)^m (a + b x^2 + c x^4)^p dx \rightarrow \int \text{ExpandIntegrand}[(dx)^m (a + b x^2 + c x^4)^p, x] dx$$

Program code:

```
Int[(d.*x_)^m.*(a+b.*x^2+c.*x^4)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d*x)^m*(a+b*x^2+c*x^4)^p,x],x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[p,0] && Not[IntegerQ[(m+1)/2]]
```

3. $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac = 0$

x: $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac = 0 \wedge p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2 - 4ac = 0$, then $a+bx^2+cx^4 = \frac{1}{c} \left(\frac{b}{2} + cx^2\right)^2$

Rule 1.2.2.2.3.1: If $b^2 - 4ac = 0 \wedge p \in \mathbb{Z}$, then

$$\int (dx)^m (a+bx^2+cx^4)^p dx \rightarrow \frac{1}{c^p} \int (dx)^m \left(\frac{b}{2} + cx^2\right)^{2p} dx$$

Program code:

```
(* Int[(d_.x_)^m_.*(a+b_.x_^2+c_.x_^4)^p_,x_Symbol] :=
  1/c^p*Int[(d*x)^m*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p] *)
```

$$2. \int (d x)^m (a + b x^2 + c x^4)^p dx \text{ when } b^2 - 4 a c == 0 \wedge p \notin \mathbb{Z}$$

$$1. \int (d x)^m (a + b x^2 + c x^4)^p dx \text{ when } b^2 - 4 a c == 0 \wedge p \notin \mathbb{Z} \wedge m + 4 p + 5 == 0 \wedge p \neq -\frac{1}{2}$$

$$1: \int (d x)^m (a + b x^2 + c x^4)^p dx \text{ when } b^2 - 4 a c == 0 \wedge p \notin \mathbb{Z} \wedge m + 4 p + 5 == 0 \wedge p < -1$$

Derivation: Square trinomial recurrence 2c with $m + 4 p + 5 == 0$

Rule 1.2.2.2.3.2.1: If $b^2 - 4 a c == 0 \wedge p \notin \mathbb{Z} \wedge m + 4 p + 5 == 0 \wedge p < -1$, then

$$\int (d x)^m (a + b x^2 + c x^4)^p dx \rightarrow \frac{2 (d x)^{m+1} (a + b x^2 + c x^4)^{p+1}}{d (m+3) (2 a + b x^2)} - \frac{(d x)^{m+1} (a + b x^2 + c x^4)^{p+1}}{2 a d (m+3) (p+1)}$$

Program code:

```
Int[(d_.**x_)^m_.*(a_+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  2*(d*x)^(m+1)*(a+b*x^2+c*x^4)^(p+1)/(d*(m+3)*(2*a+b*x^2)) -
  (d*x)^(m+1)*(a+b*x^2+c*x^4)^(p+1)/(2*a*d*(m+3)*(p+1)) /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && EqQ[m+4*p+5,0] && LtQ[p,-1]
```

$$\text{2: } \int (dx)^m (a+bx^2+cx^4)^p dx \text{ when } b^2 - 4ac = 0 \wedge p \notin \mathbb{Z} \wedge m+4p+5 = 0 \wedge p \neq -\frac{1}{2}$$

Derivation: Square trinomial recurrence 2c with $m+4p+5 = 0$

Rule 1.2.2.2.3.2.1: If $b^2 - 4ac = 0 \wedge p \notin \mathbb{Z} \wedge m+4p+5 = 0 \wedge p \neq -\frac{1}{2}$, then

$$\int (dx)^m (a+bx^2+cx^4)^p dx \rightarrow \frac{(dx)^{m+1} (a+bx^2+cx^4)^{p+1}}{4ad(p+1)(2p+1)} - \frac{(dx)^{m+1} (2a+bx^2) (a+bx^2+cx^4)^p}{4ad(2p+1)}$$

Program code:

```
Int[(d_.**x_)^m_.*(a_+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^2+c*x^4)^(p+1)/(4*a*d*(p+1)*(2*p+1)) -
  (d*x)^(m+1)*(2*a+b*x^2)*(a+b*x^2+c*x^4)^p/(4*a*d*(2*p+1)) /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && EqQ[m+4*p+5,0] && NeQ[p,-1/2]
```

$$\text{?: } \int x^m (a+bx^2+cx^4)^p dx \text{ when } b^2 - 4ac = 0 \wedge p - \frac{1}{2} \in \mathbb{Z} \wedge \frac{m-1}{2} \in \mathbb{Z}$$

Derivation: Integration by substitution

Basis: If $\frac{m-1}{2} \in \mathbb{Z}$, then $x^m F[x^2] = \frac{1}{2} \text{Subst}[x^{\frac{m-1}{2}} F[x], x, x^2] \partial_x x^2$

Rule 1.2.2.2.5.1: If $b^2 - 4ac = 0 \wedge p - \frac{1}{2} \in \mathbb{Z} \wedge \frac{m-1}{2} \in \mathbb{Z}$, then

$$\int x^m (a+bx^2+cx^4)^p dx \rightarrow \frac{1}{2} \text{Subst}\left[\int x^{\frac{m-1}{2}} (a+bx+cx^2)^p dx, x, x^2\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  1/2*Subst[Int[x^((m-1)/2)*(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p-1/2] && IntegerQ[(m-1)/2] && (GtQ[m,0] || LtQ[0,4*p,-m-1])
```

2: $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac = 0 \wedge p - \frac{1}{2} \in \mathbb{Z} \wedge m \in \mathbb{Z}^+$ **Delete !**

Derivation: Piecewise constant extraction

Basis: If $b^2 - 4ac = 0$, then $\partial_x \frac{(a+bx^2+cx^4)^{p+1}}{\left(\frac{b}{2}+cx^2\right)^{2(p+1)}} = 0$

Rule 1.2.2.2.3.2.2: If $b^2 - 4ac = 0 \wedge p - \frac{1}{2} \in \mathbb{Z} \wedge m \in \mathbb{Z}^+$, then

$$\int (dx)^m (a+bx^2+cx^4)^p dx \rightarrow \frac{c(a+bx^2+cx^4)^{p+1}}{\left(\frac{b}{2}+cx^2\right)^{2(p+1)}} \int (dx)^m \left(\frac{b}{2}+cx^2\right)^{2p} dx$$

Program code:

```
(* Int[(d_.*x_)^m_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  c*(a+b*x^2+c*x^4)^(p+1)/(b/2+c*x^2)^(2*(p+1))*Int[(d*x)^m*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p-1/2] && IGeQ[m,2*p] *)
```

$$\mathbf{3:} \int (dx)^m (a+bx^2+cx^4)^p dx \text{ when } b^2 - 4ac = 0 \wedge p \notin \mathbb{Z}$$

Derivation: Piecewise constant extraction

$$\text{Basis: If } b^2 - 4ac = 0, \text{ then } \partial_x \frac{(a+bx^2+cx^4)^p}{\left(\frac{b}{2}+cx^2\right)^{2p}} = 0$$

$$\text{Note: If } b^2 - 4ac = 0, \text{ then } a+bx^2+cx^4 = \frac{1}{c} \left(\frac{b}{2}+cx^2\right)^2$$

Rule 1.2.2.2.3.2.2: If $b^2 - 4ac = 0 \wedge p \notin \mathbb{Z}$, then

$$\int (dx)^m (a+bx^2+cx^4)^p dx \rightarrow \frac{(a+bx^2+cx^4)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]} \left(\frac{b}{2}+cx^2\right)^{2\text{FracPart}[p]}} \int (dx)^m \left(\frac{b}{2}+cx^2\right)^{2p} dx$$

Program code:

```
Int[(d_.x_)^m_.*(a+b_.x_^2+c_.x_^4)^p_,x_Symbol] :=
  (a+b*x^2+c*x^4)^FracPart[p]/(c^IntPart[p]*(b/2+c*x^2)^(2*FracPart[p]))*Int[(d*x)^m*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p-1/2]
```

```
Int[(d_.x_)^m_.*(a+b_.x_^2+c_.x_^4)^p_,x_Symbol] :=
  a^IntPart[p]*(a+b*x^2+c*x^4)^FracPart[p]/(1+2*c*x^2/b)^(2*FracPart[p])*Int[(d*x)^m*(1+2*c*x^2/b)^(2*p),x] /;
FreeQ[{a,b,c,d,m,p},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[2*p]]
```


4: $\int x^m (a + b x^2 + c x^4)^p dx$ when $\frac{m-1}{2} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m-1}{2} \in \mathbb{Z}$, then $x^m F[x^2] = \frac{1}{2} \text{Subst}\left[x^{\frac{m-1}{2}} F[x], x, x^2\right] \partial_x x^2$

Rule 1.2.2.2.5.1: If $\frac{m-1}{2} \in \mathbb{Z}$, then

$$\int x^m (a + b x^2 + c x^4)^p dx \rightarrow \frac{1}{2} \text{Subst}\left[\int x^{\frac{m-1}{2}} (a + b x + c x^2)^p dx, x, x^2\right]$$

Program code:

```
Int[x_^m.*(a_+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  1/2*Subst[Int[x^((m-1)/2)*(a+b*x+c*x^2)^p,x],x,x^2] /;
FreeQ[{a,b,c,p},x] && IntegerQ[(m-1)/2]
```

5: $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac \neq 0 \wedge m \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $(dx)^m F[x] = \frac{k}{d} \text{Subst}\left[x^{k(m+1)-1} F\left[\frac{x^k}{d}\right], x, (dx)^{1/k}\right] \partial_x (dx)^{1/k}$

Rule 1.2.2.6.1.2: If $b^2 - 4ac \neq 0 \wedge m \in \mathbb{F}$, let $k = \text{Denominator}[m]$, then

$$\int (dx)^m (a+bx^2+cx^4)^p dx \rightarrow \frac{k}{d} \text{Subst}\left[\int x^{k(m+1)-1} \left(a + \frac{bx^{2k}}{d^2} + \frac{cx^{4k}}{d^4}\right)^p dx, x, (dx)^{1/k}\right]$$

Program code:

```
Int[(d_.**x_)^m_*(a_+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  With[{k=Denominator[m]},
    k/d*Subst[Int[x^(k*(m+1)-1)*(a+b*x^(2*k)/d^2+c*x^(4*k)/d^4)^p,x],x,(d*x)^(1/k)]] /;
  FreeQ[{a,b,c,d,p},x] && NeQ[b^2-4*a*c,0] && FractionQ[m] && IntegerQ[p]
```

6. $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac \neq 0 \wedge p > 0$

1: $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac \neq 0 \wedge p > 0 \wedge m > 1$

Derivation: Trinomial recurrence 1b with $A = 0$, $B = 1$ and $m = m - n$

Rule 1.2.2.2.6.1.3.1: If $b^2 - 4ac \neq 0 \wedge p > 0 \wedge m > 1$, then

$$\int (dx)^m (a+bx^2+cx^4)^p dx \rightarrow \frac{d (dx)^{m-1} (a+bx^2+cx^4)^p (2bp + c(m+4p-1)x^2)}{c(m+4p+1)(m+4p-1)} - \frac{2pd^2}{c(m+4p+1)(m+4p-1)} \int (dx)^{m-2} (a+bx^2+cx^4)^{p-1} (ab(m-1) - (2ac(m+4p-1) - b^2(m+2p-1))x^2) dx$$

Program code:

```
Int[(d_.**x_)^m_.*(a_+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  d*(d*x)^(m-1)*(a+b*x^2+c*x^4)^p*(2*b*p+c*(m+4*p-1)*x^2)/(c*(m+4*p+1)*(m+4*p-1)) -
  2*p*d^2/(c*(m+4*p+1)*(m+4*p-1))*
  Int[(d*x)^(m-2)*(a+b*x^2+c*x^4)^(p-1)*Simp[a*b*(m-1)-(2*a*c*(m+4*p-1)-b^2*(m+2*p-1))*x^2,x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && GtQ[m,1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

2: $\int (d x)^m (a + b x^2 + c x^4)^p dx$ when $b^2 - 4 a c \neq 0 \wedge p > 0 \wedge m < -1$

Reference: G&R 2.160.2

Derivation: Trinomial recurrence 1a with $A = 1$ and $B = 0$

Rule 1.2.2.2.6.1.3.2: If $b^2 - 4 a c \neq 0 \wedge p > 0 \wedge m < -1$, then

$$\int (d x)^m (a + b x^2 + c x^4)^p dx \rightarrow \frac{(d x)^{m+1} (a + b x^2 + c x^4)^p}{d (m+1)} - \frac{2 p}{d^2 (m+1)} \int (d x)^{m+2} (b + 2 c x^2) (a + b x^2 + c x^4)^{p-1} dx$$

Program code:

```
Int[(d_.**x_)^m_.*(a_+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^2+c*x^4)^p/(d*(m+1)) -
  2*p/(d^2*(m+1))*Int[(d*x)^(m+2)*(b+2*c*x^2)*(a+b*x^2+c*x^4)^(p-1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && LtQ[m,-1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

3: $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac \neq 0 \wedge p > 0 \wedge m+4p+1 \neq 0$

Derivation: Trinomial recurrence 1a with $A = 0, B = 1$ and $m = m - n$

Derivation: Trinomial recurrence 1b with $A = 1$ and $B = 0$

Rule 1.2.2.2.6.1.3.4: If $b^2 - 4ac \neq 0 \wedge p > 0 \wedge m+4p+1 \neq 0$, then

$$\int (dx)^m (a+bx^2+cx^4)^p dx \rightarrow \frac{(dx)^{m+1} (a+bx^2+cx^4)^p}{d(m+4p+1)} + \frac{2p}{m+4p+1} \int (dx)^m (2a+bx^2) (a+bx^2+cx^4)^{p-1} dx$$

Program code:

```
Int[(d_.**x_)^m_.*(a_+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^2+c*x^4)^p/(d*(m+4*p+1)) +
  2*p/(m+4*p+1)*Int[(d*x)^m*(2*a+b*x^2)*(a+b*x^2+c*x^4)^(p-1),x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[b^2-4*a*c,0] && GtQ[p,0] && NeQ[m+4*p+1,0] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

7. $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac \neq 0 \wedge p < -1$

1. $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac \neq 0 \wedge p < -1 \wedge m > 1$

1: $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac \neq 0 \wedge p < -1 \wedge 1 < m \leq 3$

Derivation: Trinomial recurrence 2a with $A = 1$ and $B = 0$

Derivation: Trinomial recurrence 2b with $A = 0, B = 1$ and $m = m - n$

Rule 1.2.2.2.6.1.4.1.1: If $b^2 - 4ac \neq 0 \wedge p < -1 \wedge 1 < m \leq 3$, then

$$\int (dx)^m (a+bx^2+cx^4)^p dx \rightarrow$$

$$\frac{d (d x)^{m-1} (b+2 c x^2) (a+b x^2+c x^4)^{p+1}}{2 (p+1) (b^2-4 a c)} - \frac{d^2}{2 (p+1) (b^2-4 a c)} \int (d x)^{m-2} (b (m-1)+2 c (m+4 p+5) x^2) (a+b x^2+c x^4)^{p+1} dx$$

Program code:

```
Int[(d_.**x_)^m_.*(a_+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  d*(d*x)^(m-1)*(b+2*c*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*(p+1)*(b^2-4*a*c)) -
  d^2/(2*(p+1)*(b^2-4*a*c))*Int[(d*x)^(m-2)*(b*(m-1)+2*c*(m+4*p+5)*x^2)*(a+b*x^2+c*x^4)^(p+1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && GtQ[m,1] && LeQ[m,3] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

2: $\int (d x)^m (a+b x^2+c x^4)^p dx$ when $b^2-4 a c \neq 0 \wedge p < -1 \wedge m > 3$

Derivation: Trinomial recurrence 2a with $A = 0, B = 1$ and $m = m - n$

Rule 1.2.2.2.6.1.4.1.2: If $b^2-4 a c \neq 0 \wedge p < -1 \wedge m > 3$, then

$$\int (d x)^m (a+b x^2+c x^4)^p dx \rightarrow -\frac{d^3 (d x)^{m-3} (2 a+b x^2) (a+b x^2+c x^4)^{p+1}}{2 (p+1) (b^2-4 a c)} + \frac{d^4}{2 (p+1) (b^2-4 a c)} \int (d x)^{m-4} (2 a (m-3)+b (m+4 p+3) x^2) (a+b x^2+c x^4)^{p+1} dx$$

Program code:

```
Int[(d_.**x_)^m_.*(a_+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  -d^3*(d*x)^(m-3)*(2*a+b*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*(p+1)*(b^2-4*a*c)) +
  d^4/(2*(p+1)*(b^2-4*a*c))*Int[(d*x)^(m-4)*(2*a*(m-3)+b*(m+4*p+3)*x^2)*(a+b*x^2+c*x^4)^(p+1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && GtQ[m,3] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

2: $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac \neq 0 \wedge p < -1$

Derivation: Trinomial recurrence 2b with $A = 1$ and $B = 0$

Rule 1.2.2.2.6.1.4.2: If $b^2 - 4ac \neq 0 \wedge p < -1$, then

$$\int (dx)^m (a+bx^2+cx^4)^p dx \rightarrow -\frac{(dx)^{m+1} (b^2 - 2ac + bcx^2) (a+bx^2+cx^4)^{p+1}}{2ad(p+1)(b^2 - 4ac)} + \frac{1}{2a(p+1)(b^2 - 4ac)} \int (dx)^m (a+bx^2+cx^4)^{p+1} (b^2(m+2p+3) - 2ac(m+4p+5) + bc(m+4p+7)x^2) dx$$

Program code:

```
Int[(d.*x_)^m.*(a+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  -(d*x)^(m+1)*(b^2-2*a*c+b*c*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*a*d*(p+1)*(b^2-4*a*c)) +
  1/(2*a*(p+1)*(b^2-4*a*c))*
  Int[(d*x)^m*(a+b*x^2+c*x^4)^(p+1)*Simp[b^2*(m+2*p+3)-2*a*c*(m+4*p+5)+b*c*(m+4*p+7)*x^2,x],x] /;
FreeQ[{a,b,c,d,m},x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

8: $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac \neq 0 \wedge m > 3 \wedge m+4p+1 \neq 0$

Reference: G&R 2.160.3

Derivation: Trinomial recurrence 3a with $A = 0$, $B = 1$ and $m = m - n$

Note: G&R 2.174.1 is a special case of G&R 2.160.3.

Rule 1.2.2.2.6.1.5: If $b^2 - 4ac \neq 0 \wedge m > 3 \wedge m+4p+1 \neq 0$, then

$$\int (dx)^m (a+bx^2+cx^4)^p dx \rightarrow$$

$$\frac{d^3 (dx)^{m-3} (a+bx^2+cx^4)^{p+1}}{c(m+4p+1)} - \frac{d^4}{c(m+4p+1)} \int (dx)^{m-4} (a(m-3) + b(m+2p-1)x^2) (a+bx^2+cx^4)^p dx$$

Program code:

```
Int[(d_.**x_)^m_.*(a_+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  d^3*(d*x)^(m-3)*(a+b*x^2+c*x^4)^(p+1)/(c*(m+4*p+1)) -
  d^4/(c*(m+4*p+1))*
  Int[(d*x)^(m-4)*Simp[a*(m-3)+b*(m+2*p-1)*x^2,x]*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,p},x] && NeQ[b^2-4*a*c,0] && GtQ[m,3] && NeQ[m+4*p+1,0] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```

9: $\int (dx)^m (a+bx^2+cx^4)^p dx$ when $b^2 - 4ac \neq 0 \wedge m < -1$

Reference: G&R 2.160.1

Derivation: Trinomial recurrence 3b with $A = 1$ and $B = 0$

Note: G&R 2.161.6 is a special case of G&R 2.160.1.

Rule 1.2.2.2.6.1.6: If $b^2 - 4ac \neq 0 \wedge m < -1$, then

$$\int (dx)^m (a+bx^2+cx^4)^p dx \rightarrow \frac{(dx)^{m+1} (a+bx^2+cx^4)^{p+1}}{ad(m+1)} - \frac{1}{ad^2(m+1)} \int (dx)^{m+2} (b(m+2p+3) + c(m+4p+5)x^2) (a+bx^2+cx^4)^p dx$$

Program code:

```
Int[(d_.**x_)^m_.*(a_+b_.**x_^2+c_.**x_^4)^p_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*x^2+c*x^4)^(p+1)/(a*d*(m+1)) -
  1/(a*d^2*(m+1))*Int[(d*x)^(m+2)*(b*(m+2*p+3)+c*(m+4*p+5)*x^2)*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,p},x] && NeQ[b^2-4*a*c,0] && LtQ[m,-1] && IntegerQ[2*p] && (IntegerQ[p] || IntegerQ[m])
```


$$10. \int \frac{(dx)^m}{a+bx^2+cx^4} dx \text{ when } b^2 - 4ac \neq 0$$

$$1: \int \frac{(dx)^m}{a+bx^2+cx^4} dx \text{ when } b^2 - 4ac \neq 0 \wedge m < -1$$

Reference: G&R 2.176, CRC 123

Derivation: Algebraic expansion

$$\text{Basis: } \frac{(dz)^m}{a+bz+cz^2} = \frac{(dz)^m}{a} - \frac{1}{ad} \frac{(dz)^{m+1}(b+cz)}{a+bz+cz^2}$$

Rule 1.2.2.2.6.1.7.1: If $b^2 - 4ac \neq 0 \wedge m < -1$, then

$$\int \frac{(dx)^m}{a+bx^2+cx^4} dx \rightarrow \frac{(dx)^{m+1}}{ad(m+1)} - \frac{1}{ad^2} \int \frac{(dx)^{m+2}(b+cx^2)}{a+bx^2+cx^4} dx$$

Program code:

```
Int[(d.*x_)^m/(a+b.*x_^2+c.*x_^4),x_Symbol] :=
  (d*x)^(m+1)/(a*d*(m+1)) -
  1/(a*d^2)*Int[(d*x)^(m+2)*(b+c*x^2)/(a+b*x^2+c*x^4),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && LtQ[m,-1]
```

$$2. \int \frac{(dx)^m}{a+bx^2+cx^4} dx \text{ when } b^2 - 4ac \neq 0 \wedge m > 3$$

$$1: \int \frac{x^m}{a+bx^2+cx^4} dx \text{ when } b^2 - 4ac \neq 0 \wedge m > 5 \wedge m \in \mathbb{Z}$$

Derivation: Algebraic expansion

Rule 1.2.2.2.6.1.7.2.1: If $b^2 - 4ac \neq 0 \wedge m > 5 \wedge m \in \mathbb{Z}$, then

$$\int \frac{x^m}{a+bx^2+cx^4} dx \rightarrow \int \text{PolynomialDivide}[x^m, a+bx^2+cx^4, x] dx$$

Program code:

```
Int[x^m/(a+b*x^2+c*x^4),x_Symbol] :=
  Int[PolynomialDivide[x^m,(a+b*x^2+c*x^4),x],x] /;
FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && IGtQ[m,5]
```

2: $\int \frac{(dx)^m}{a+bx^2+cx^4} dx$ when $b^2 - 4ac \neq 0 \wedge m > 3$ **Not necessary?**

Reference: G&R 2.174.1, CRC 119

Derivation: Algebraic expansion

Basis: $\frac{(dz)^m}{a+bz+cz^2} = \frac{d^2 (dz)^{m-2}}{c} - \frac{d^2}{c} \frac{(dz)^{m-2} (a+bz)}{a+bz+cz^2}$

Rule 1.2.2.2.6.1.7.2.2: If $b^2 - 4ac \neq 0 \wedge m > 3$, then

$$\int \frac{(dx)^m}{a+bx^2+cx^4} dx \rightarrow \frac{d^3 (dx)^{m-3}}{c(m-3)} - \frac{d^4}{c} \int \frac{(dx)^{m-4} (a+bx^2)}{a+bx^2+cx^4} dx$$

Program code:

```
Int[(d.*x_)^m/(a+b.*x_^2+c.*x_^4),x_Symbol] :=
  d^3*(d*x)^(m-3)/(c*(m-3)) - d^4/c*Int[(d*x)^(m-4)*(a+b*x^2)/(a+b*x^2+c*x^4),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && GtQ[m,3]
```

$$3. \int \frac{x^m}{a+bx^2+cx^4} dx \text{ when } b^2 - 4ac \neq 0 \wedge m \in \mathbb{Z}^+ \wedge 1 \leq m < 4 \wedge b^2 - 4ac \neq 0$$

$$1: \int \frac{x^2}{a+bx^2+cx^4} dx \text{ when } b^2 - 4ac < 0 \wedge ac > 0$$

Derivation: Algebraic expansion

Basis: Let $q \rightarrow \sqrt{\frac{a}{c}}$, then $\frac{x^2}{a+bx^2+cx^4} = \frac{q+x^2}{2(a+bx^2+cx^4)} - \frac{q-x^2}{2(a+bx^2+cx^4)}$

Note: Resulting integrands are of the form $\frac{d+ex^2}{a+bx^2+cx^4}$ where $c d^2 - a e^2 = 0 \wedge b^2 - 4ac \neq 0$, for which there is rule.

Rule 1.2.2.2.6.1.7.3.1: If $b^2 - 4ac < 0 \wedge ac > 0$, let $q \rightarrow \sqrt{\frac{a}{c}}$, then

$$\int \frac{x^2}{a+bx^2+cx^4} dx \rightarrow \frac{1}{2} \int \frac{q+x^2}{a+bx^2+cx^4} dx - \frac{1}{2} \int \frac{q-x^2}{a+bx^2+cx^4} dx$$

Program code:

```
Int[x^2/(a+b.*x^2+c.*x^4), x_Symbol] :=
  With[{q=Rt[a/c,2]},
    1/2*Int[(q+x^2)/(a+b*x^2+c*x^4),x] - 1/2*Int[(q-x^2)/(a+b*x^2+c*x^4),x] /;
    FreeQ[{a,b,c},x] && LtQ[b^2-4*a*c,0] && PosQ[a*c]
```

2: $\int \frac{x^m}{a+bx^2+cx^4} dx$ when $b^2 - 4ac \neq 0 \wedge m \in \mathbb{Z}^+ \wedge 3 \leq m < 4 \wedge b^2 - 4ac \neq 0$

Derivation: Algebraic expansion

Basis: If $q \rightarrow \sqrt{\frac{a}{c}}$ and $r \rightarrow \sqrt{2q - \frac{b}{c}}$, then $\frac{z^3}{a+bz^2+cz^4} = \frac{q+rz}{2cr(q+rz+z^2)} - \frac{q-rz}{2cr(q-rz+z^2)}$

Note: If $(a | b | c) \in \mathbb{R} \wedge b^2 - 4ac < 0$, then $\frac{a}{c} > 0$ and $2\sqrt{\frac{a}{c} - \frac{b}{c}} > 0$.

Rule 1.2.2.2.6.1.7.3.2: If $b^2 - 4ac \neq 0 \wedge m \in \mathbb{Z}^+ \wedge 3 \leq m < 4 \wedge b^2 - 4ac \neq 0$, let $q \rightarrow \sqrt{\frac{a}{c}}$ and $r \rightarrow \sqrt{2q - \frac{b}{c}}$, then

$$\int \frac{x^m}{a+bx^2+cx^4} dx \rightarrow \frac{1}{2cr} \int \frac{x^{m-3}(q+rx)}{q+rx+x^2} dx - \frac{1}{2cr} \int \frac{x^{m-3}(q-rx)}{q-rx+x^2} dx$$

Program code:

```
Int[x_^m_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[a/c,2]},
    With[{r=Rt[2*q-b/c,2]},
      1/(2*c*r)*Int[x^(m-3)*(q+r*x)/(q+r*x+x^2),x] -
      1/(2*c*r)*Int[x^(m-3)*(q-r*x)/(q-r*x+x^2),x]] /;
  FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && GeQ[m,3] && LtQ[m,4] && NegQ[b^2-4*a*c]
```

3: $\int \frac{x^m}{a+bx^2+cx^4} dx$ when $b^2 - 4ac \neq 0 \wedge m \in \mathbb{Z}^+ \wedge 1 \leq m < 3 \wedge b^2 - 4ac \neq 0$

Derivation: Algebraic expansion

Basis: If $q \rightarrow \sqrt{\frac{a}{c}}$ and $r \rightarrow \sqrt{2q - \frac{b}{c}}$, then $\frac{z}{a+bz^2+cz^4} = \frac{1}{2cr(q-rz+z^2)} - \frac{1}{2cr(q+rz+z^2)}$

Note: If $(a | b | c) \in \mathbb{R} \wedge b^2 - 4ac < 0$, then $\frac{a}{c} > 0$ and $2\sqrt{\frac{a}{c}} - \frac{b}{c} > 0$.

Rule 1.2.2.2.6.1.7.3.3: If $b^2 - 4ac \neq 0 \wedge m \in \mathbb{Z}^+ \wedge 1 \leq m < 3 \wedge b^2 - 4ac \neq 0$, let $q \rightarrow \sqrt{\frac{a}{c}}$ and $r \rightarrow \sqrt{2q - \frac{b}{c}}$, then

$$\int \frac{x^m}{a+bx^2+cx^4} dx \rightarrow \frac{1}{2cr} \int \frac{x^{m-1}}{q-rx+x^2} dx - \frac{1}{2cr} \int \frac{x^{m-1}}{q+rx+x^2} dx$$

Program code:

```
Int[x^m_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[a/c,2]},
    With[{r=Rt[2*q-b/c,2]},
      1/(2*c*r)*Int[x^(m-1)/(q-r*x+x^2),x] - 1/(2*c*r)*Int[x^(m-1)/(q+r*x+x^2),x]] /;
  FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && GeQ[m,1] && LtQ[m,3] && NegQ[b^2-4*a*c]
```

4: $\int \frac{(dx)^m}{a+bx^2+cx^4} dx$ when $b^2 - 4ac \neq 0 \wedge m \geq 2$

Reference: G&R 2.161.1a & G&R 2.161.3

Derivation: Algebraic expansion

Basis: Let $q \rightarrow \sqrt{b^2 - 4ac}$, then $\frac{(dz)^m}{a+bz+cz^2} = \frac{d}{2} \left(\frac{b}{q} + 1 \right) \frac{(dz)^{m-1}}{\frac{b}{2} + \frac{q}{2} + cz} - \frac{d}{2} \left(\frac{b}{q} - 1 \right) \frac{(dz)^{m-1}}{\frac{b}{2} - \frac{q}{2} + cz}$

■ Rule 1.2.2.2.6.1.7.4: If $b^2 - 4ac \neq 0 \wedge m \geq 2$, let $q \rightarrow \sqrt{b^2 - 4ac}$, then

$$\int \frac{(dx)^m}{a+bx^2+cx^4} dx \rightarrow \frac{d^2}{2} \left(\frac{b}{q} + 1 \right) \int \frac{(dx)^{m-2}}{\frac{b}{2} + \frac{q}{2} + cx^2} dx - \frac{d^2}{2} \left(\frac{b}{q} - 1 \right) \int \frac{(dx)^{m-2}}{\frac{b}{2} - \frac{q}{2} + cx^2} dx$$

Program code:

```
Int[(d.*x_)^m/(a+b.*x_^2+c.*x_^4),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    d^2/2*(b/q+1)*Int[(d*x)^(m-2)/(b/2+q/2+c*x^2),x] -
    d^2/2*(b/q-1)*Int[(d*x)^(m-2)/(b/2-q/2+c*x^2),x] /;
  FreeQ[{a,b,c,d},x] && NeQ[b^2-4*a*c,0] && GeQ[m,2]
```

5: $\int \frac{(dx)^m}{a+bx^2+cx^4} dx$ when $b^2 - 4ac \neq 0$

Reference: G&R 2.161.1a

Derivation: Algebraic expansion

Basis: Let $q \rightarrow \sqrt{b^2 - 4ac}$, then $\frac{1}{a+bz+cz^2} = \frac{c}{q} \frac{1}{\frac{b}{2} - \frac{q}{2} + cz} - \frac{c}{q} \frac{1}{\frac{b}{2} + \frac{q}{2} + cz}$

■ Rule 1.2.2.2.6.1.7.5: If $b^2 - 4ac \neq 0$, let $q \rightarrow \sqrt{b^2 - 4ac}$, then

$$\int \frac{(dx)^m}{a+bx^2+cx^4} dx \rightarrow \frac{c}{q} \int \frac{(dx)^m}{\frac{b}{2} - \frac{q}{2} + cx^2} dx - \frac{c}{q} \int \frac{(dx)^m}{\frac{b}{2} + \frac{q}{2} + cx^2} dx$$

Program code:

```
Int[(d_.**x_)^m_./(a_+b_.**x_^2+c_.**x_^4),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    c/q*Int[(d*x)^m/(b/2-q/2+c*x^2),x] - c/q*Int[(d*x)^m/(b/2+q/2+c*x^2),x] /;
  FreeQ[{a,b,c,d,m},x] && NeQ[b^2-4*a*c,0]
```

11. $\int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx$ when $b^2 - 4ac \neq 0$

1. $\int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx$ when $b^2 - 4ac > 0$

1: $\int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx$ when $b^2 - 4ac > 0 \wedge c < 0$

Derivation: Algebraic expansion

Basis: If $b^2 - 4ac > 0 \wedge c < 0$, let $q \rightarrow \sqrt{b^2 - 4ac}$, then

$$\sqrt{a+bx^2+cx^4} = \frac{1}{2\sqrt{-c}} \sqrt{b+q+2cx^2} \sqrt{-b+q-2cx^2}$$

■ Rule 1.2.2.2.6.1.8.1.1: If $b^2 - 4ac > 0 \wedge c < 0$, let $q \rightarrow \sqrt{b^2 - 4ac}$, then

$$\int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \rightarrow 2\sqrt{-c} \int \frac{x^2}{\sqrt{b+q+2cx^2} \sqrt{-b+q-2cx^2}} dx$$

Program code:

```
Int[x^2/Sqrt[a+b_.*x^2+c_.*x^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    2*Sqrt[-c]*Int[x^2/(Sqrt[b+q+2*c*x^2]*Sqrt[-b+q-2*c*x^2]),x] /;
  FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0] && LtQ[c,0]
```

$$2. \int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \text{ when } b^2 - 4ac > 0 \wedge c \neq 0$$

$$1: \int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \text{ when } b^2 - 4ac > 0 \wedge \frac{c}{a} > 0 \wedge \frac{b}{a} < 0$$

Derivation: Algebraic expansion

■ Rule 1.2.2.2.6.1.8.1.2.1: If $b^2 - 4ac > 0 \wedge \frac{c}{a} > 0 \wedge \frac{b}{a} < 0$, let $q \rightarrow \sqrt{\frac{c}{a}}$, then

$$\int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \rightarrow \frac{1}{q} \int \frac{1}{\sqrt{a+bx^2+cx^4}} dx - \frac{1}{q} \int \frac{1-qx^2}{\sqrt{a+bx^2+cx^4}} dx$$

Program code:

```
Int[x^2/Sqrt[a+b_.*x^2+c_.*x^4],x_Symbol] :=
  With[{q=Rt[c/a,2]},
    1/q*Int[1/Sqrt[a+b*x^2+c*x^4],x] - 1/q*Int[(1-q*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
  FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0] && GtQ[c/a,0] && LtQ[b/a,0]
```

2: $\int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx$ when $b^2 - 4ac > 0 \wedge a < 0 \wedge c > 0$

Derivation: Algebraic expansion

■ Rule 1.2.2.2.6.1.8.1.2.2: If $b^2 - 4ac > 0 \wedge a < 0 \wedge c > 0$, let $q \rightarrow \sqrt{b^2 - 4ac}$, then

$$\int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \rightarrow -\frac{b-q}{2c} \int \frac{1}{\sqrt{a+bx^2+cx^4}} dx + \frac{1}{2c} \int \frac{b-q+2cx^2}{\sqrt{a+bx^2+cx^4}} dx$$

Program code:

```
Int[x_^2/Sqrt[a+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    -(b-q)/(2*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + 1/(2*c)*Int[(b-q+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
    FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0] && LtQ[a,0] && GtQ[c,0]
```

$$3. \int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \text{ when } b^2 - 4ac > 0 \wedge \frac{b+\sqrt{b^2-4ac}}{a} > 0$$

$$1: \int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \text{ when } b^2 - 4ac > 0 \wedge \frac{b+\sqrt{b^2-4ac}}{a} > 0$$

Reference: G&R 3.153.1+

Rule 1.2.2.2.6.1.8.1.2.3.1: If $b^2 - 4ac > 0$, let $q \rightarrow \sqrt{b^2 - 4ac}$, if $\frac{b+q}{a} > 0$, then

$$\int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \rightarrow \frac{x(b+q+2cx^2)}{2c\sqrt{a+bx^2+cx^4}} - \frac{\sqrt{\frac{b+q}{2a}}(2a+(b+q)x^2)\sqrt{\frac{2a+(b-q)x^2}{2a+(b+q)x^2}}}{2c\sqrt{a+bx^2+cx^4}} \text{EllipticE}\left[\text{ArcTan}\left[\sqrt{\frac{b+q}{2a}}x\right], \frac{2q}{b+q}\right]$$

Program code:

```
Int[x^2/Sqrt[a+b_.*x^2+c_.*x^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    x*(b+q+2*c*x^2)/(2*c*Sqrt[a+b*x^2+c*x^4]) -
    Rt[(b+q)/(2*a),2]*(2*a+(b+q)*x^2)*Sqrt[(2*a+(b-q)*x^2)/(2*a+(b+q)*x^2)]/(2*c*Sqrt[a+b*x^2+c*x^4])*
    EllipticE[ArcTan[Rt[(b+q)/(2*a),2]*x],2*q/(b+q)] /;
    PosQ[(b+q)/a] && Not[PosQ[(b-q)/a] && SimplifySqrtQ[(b-q)/(2*a),(b+q)/(2*a)]] /;
    FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

$$2: \int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \text{ when } b^2 - 4ac > 0 \wedge \frac{b-\sqrt{b^2-4ac}}{a} > 0$$

Reference: G&R 3.153.1-

Rule 1.2.2.2.6.1.8.1.2.3.2: If $b^2 - 4ac > 0$, let $q \rightarrow \sqrt{b^2 - 4ac}$, if $\frac{b-q}{a} > 0$ then

$$\int \frac{x^2}{\sqrt{a + b x^2 + c x^4}} dx \rightarrow \frac{x (b - q + 2 c x^2)}{2 c \sqrt{a + b x^2 + c x^4}} - \frac{\sqrt{\frac{b-q}{2a}} (2a + (b-q) x^2) \sqrt{\frac{2a+(b+q)x^2}{2a+(b-q)x^2}}}{2 c \sqrt{a + b x^2 + c x^4}} \text{EllipticE}\left[\text{ArcTan}\left[\sqrt{\frac{b-q}{2a}} x\right], -\frac{2q}{b-q}\right]$$

Program code:

```
Int[x_^2/Sqrt[a+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    x*(b-q+2*c*x^2)/(2*c*Sqrt[a+b*x^2+c*x^4]) -
    Rt[(b-q)/(2*a),2]*(2*a+(b-q)*x^2)*Sqrt[(2*a+(b+q)*x^2)/(2*a+(b-q)*x^2)]/(2*c*Sqrt[a+b*x^2+c*x^4])*
    EllipticE[ArcTan[Rt[(b-q)/(2*a),2]*x],-2*q/(b-q)] /;
  PosQ[(b-q)/a] /;
  FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

$$4. \int \frac{x^2}{\sqrt{a + b x^2 + c x^4}} dx \text{ when } b^2 - 4 a c > 0 \wedge \frac{b \pm \sqrt{b^2 - 4 a c}}{a} \neq 0$$

$$1: \int \frac{x^2}{\sqrt{a + b x^2 + c x^4}} dx \text{ when } b^2 - 4 a c > 0 \wedge \frac{b + \sqrt{b^2 - 4 a c}}{a} \neq 0$$

Derivation: Algebraic expansion

■ Rule 1.4.1.8.1.2.4.1: If $b^2 - 4 a c > 0$, let $q \rightarrow \sqrt{b^2 - 4 a c}$, if $\frac{b+q}{a} \neq 0$ then

$$\int \frac{x^2}{\sqrt{a + b x^2 + c x^4}} dx \rightarrow -\frac{b+q}{2c} \int \frac{1}{\sqrt{a + b x^2 + c x^4}} dx + \frac{1}{2c} \int \frac{b+q+2cx^2}{\sqrt{a + b x^2 + c x^4}} dx$$

Program code:

```
Int[x_^2/Sqrt[a+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    -(b+q)/(2*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + 1/(2*c)*Int[(b+q+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
    NegQ[(b+q)/a] && Not[NegQ[(b-q)/a] && SimplerSqrtQ[-(b-q)/(2*a),-(b+q)/(2*a)]] /;
  FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

2: $\int \frac{x^2}{\sqrt{a + b x^2 + c x^4}} dx$ when $b^2 - 4 a c > 0 \wedge \frac{b - \sqrt{b^2 - 4 a c}}{a} \neq 0$

Derivation: Algebraic expansion

■ Rule 1.4.1.8.1.2.4.2: If $b^2 - 4 a c > 0$, let $q \rightarrow \sqrt{b^2 - 4 a c}$, if $\frac{b-q}{a} \neq 0$ then

$$\int \frac{x^2}{\sqrt{a + b x^2 + c x^4}} dx \rightarrow -\frac{b-q}{2c} \int \frac{1}{\sqrt{a + b x^2 + c x^4}} dx + \frac{1}{2c} \int \frac{b-q+2cx^2}{\sqrt{a + b x^2 + c x^4}} dx$$

Program code:

```
Int[x_^2/Sqrt[a+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    -(b-q)/(2*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + 1/(2*c)*Int[(b-q+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
    NegQ[(b-q)/a] /;
    FreeQ[{a,b,c},x] && GtQ[b^2-4*a*c,0]
```

$$2. \int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \text{ when } b^2 - 4ac \neq 0$$

$$1: \int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \text{ when } b^2 - 4ac \neq 0 \wedge \frac{c}{a} > 0$$

Derivation: Algebraic expansion

Rule 1.2.2.2.6.1.8.2.1: If $b^2 - 4ac \neq 0 \wedge \frac{c}{a} > 0$, let $q \rightarrow \sqrt{\frac{c}{a}}$, then

$$\int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \rightarrow \frac{1}{q} \int \frac{1}{\sqrt{a+bx^2+cx^4}} dx - \frac{1}{q} \int \frac{1-qx^2}{\sqrt{a+bx^2+cx^4}} dx$$

Program code:

```
Int[x^2/Sqrt[a+b*x^2+c*x^4],x_Symbol] :=
  With[{q=Rt[c/a,2]},
    1/q*Int[1/Sqrt[a+b*x^2+c*x^4],x] - 1/q*Int[(1-q*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
    FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && PosQ[c/a]
```

$$2: \int \frac{x^2}{\sqrt{a+bx^2+cx^4}} dx \text{ when } b^2 - 4ac \neq 0 \wedge \frac{c}{a} \neq 0$$

Derivation: Piecewise constant extraction

$$\text{Basis: If } q \rightarrow \sqrt{b^2 - 4ac}, \text{ then } \partial_x \frac{\sqrt{1 + \frac{2cx^2}{b-q}} \sqrt{1 + \frac{2cx^2}{b+q}}}{\sqrt{a+bx^2+cx^4}} = 0$$

Rule 1.2.2.2.6.1.8.2.2: If $b^2 - 4ac \neq 0 \wedge \frac{c}{a} \neq 0$, let $q \rightarrow \sqrt{b^2 - 4ac}$, then

$$\int \frac{x^2}{\sqrt{a + b x^2 + c x^4}} dx \rightarrow \frac{\sqrt{1 + \frac{2 c x^2}{b-q}} \sqrt{1 + \frac{2 c x^2}{b+q}}}{\sqrt{a + b x^2 + c x^4}} \int \frac{x^2}{\sqrt{1 + \frac{2 c x^2}{b-q}} \sqrt{1 + \frac{2 c x^2}{b+q}}} dx$$

Program code:

```
Int[x^2/Sqrt[a+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
    Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]/Sqrt[a+b*x^2+c*x^4]*
    Int[x^2/(Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]),x] /;
  FreeQ[{a,b,c},x] && NeQ[b^2-4*a*c,0] && NegQ[c/a]
```

12: $\int (d x)^m (a + b x^2 + c x^4)^p dx$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a + b x^2 + c x^4)^p}{\left(1 + \frac{2 c x^2}{b + \sqrt{b^2 - 4 a c}}\right)^p \left(1 + \frac{2 c x^2}{b - \sqrt{b^2 - 4 a c}}\right)^p} = 0$

Rule 1.2.2.2.10:

$$\int (d x)^m (a + b x^2 + c x^4)^p dx \rightarrow \frac{a^{\text{IntPart}[p]} (a + b x^2 + c x^4)^{\text{FracPart}[p]}}{\left(1 + \frac{2 c x^2}{b + \sqrt{b^2 - 4 a c}}\right)^{\text{FracPart}[p]} \left(1 + \frac{2 c x^2}{b - \sqrt{b^2 - 4 a c}}\right)^{\text{FracPart}[p]}} \int (d x)^m \left(1 + \frac{2 c x^2}{b + \sqrt{b^2 - 4 a c}}\right)^p \left(1 + \frac{2 c x^2}{b - \sqrt{b^2 - 4 a c}}\right)^p dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  a^IntPart[p]*(a+b*x^2+c*x^4)^FracPart[p]/
  ((1+2*c*x^2/(b+Rt[b^2-4*a*c,2]))^FracPart[p]*(1+2*c*x^2/(b-Rt[b^2-4*a*c,2]))^FracPart[p])*
  Int[(d*x)^m*(1+2*c*x^2/(b+Sqrt[b^2-4*a*c]))^p*(1+2*c*x^2/(b-Sqrt[b^2-4*a*c]))^p,x] /;
  FreeQ[{a,b,c,d,m,p},x]
```

S: $\int u^m (a + b v^2 + c v^4)^p dx$ when $v = d + e x \wedge u = f v$

Derivation: Integration by substitution and piecewise constant extraction

Basis: If $u = f v$, then $\partial_x \frac{u^m}{v^m} = 0$

Rule 1.2.2.2.S: If $v = d + e x \wedge u = f v$, then

$$\int u^m (a + b v^2 + c v^4)^p dx \rightarrow \frac{u^m}{e v^m} \text{Subst} \left[\int x^m (a + b x^2 + c x^4)^p dx, x, v \right]$$

Program code:

```
Int[u_^m_.*(a_.+b_.*v_^2+c_.*v_^4)^p_,x_Symbol] :=
  u^m/(Coefficient[v,x,1]*v^m)*Subst[Int[x^m*(a+b*x^2+c*x^(2*2))^p,x],x,v] /;
FreeQ[{a,b,c,m,p},x] && LinearPairQ[u,v,x]
```