Rules for integrands involving exponentials

1. $\int u\left(F^{c\,(a+b\,x)}\right)^n\,dx$

1: $\int \left(F^{c\,(a+b\,x)}\right)^n\,dx$

Reference: G&R 2.311, CRC 519, A&S 4.2.54

Rule:

$$\int \left(F^{c\,(a+b\,x)}\right)^n\,dx \longrightarrow \frac{\left(F^{c\,(a+b\,x)}\right)^n}{b\,c\,n\,\mathrm{Log}[F]}$$

Program code:

```
Int[(F_^(c_.*(a_.+b_.*x_)))^n_.,x_Symbol] :=
  (F^(c*(a+b*x)))^n/(b*c*n*Log[F]) /;
FreeQ[{F,a,b,c,n},x]
```

2: $\int P_x\,F^{c\,v}\,dx$ when $v == a + b\,x$

Derivation: Algebraic expansion

Rule: If $v == a + b\,x$, then

$$\int P_x\,F^{c\,v}\,dx \longrightarrow \int F^{c\,(a+b\,x)}\,\mathrm{ExpandIntegrand}[P_x,\,x]\,dx$$

Program code:

```
Int[u_*F_^(c_.*v_),x_Symbol] :=
  Int[ExpandIntegrand[u*F^(c*ExpandToSum[v,x]),x],x] /;
FreeQ[{F,c},x] && PolynomialQ[u,x] && LinearQ[v,x] && $UseGamma===True
```

```
Int[u_*F_^(c_.*v_),x_Symbol] :=
  Int[ExpandIntegrand[F^(c*ExpandToSum[v,x]),u,x],x] /;
FreeQ[{F,c},x] && PolynomialQ[u,x] && LinearQ[v,x] && Not[$UseGamma===True]
```

3: $\int (d+e\,x)^m\,F^{c\,(a+b\,x)}\,(f+g\,x)\,dx$ when $e\,g\,(m+1)-b\,c\,(e\,f-d\,g)\,\text{Log}[F]==0$

Basis: $\partial_x\left(F^{f[x]}\,g[x]\right)==F^{f[x]}\,(\text{Log}[F]\,g[x]\,f'[x]+g'[x])$

Rule: If $v==a+b\,x\wedge u==d+e\,x\wedge w==f+g\,x\wedge e\,g\,(m+1)-b\,c\,(e\,f-d\,g)\,\text{Log}[F]==0$, then

$$\int u^m\,F^{c\,v}\,w\,dx\;\longrightarrow\;\int (d+e\,x)^m\,F^{c\,(a+b\,x)}\,(f+g\,x)\,dx\;\longrightarrow\;\frac{g\,(d+e\,x)^{m+1}\,F^{c\,(a+b\,x)}}{b\,c\,e\,\text{Log}[F]}$$

Program code:

```
Int[u_^m_.*F_^(c_.*v_)*w_,x_Symbol] :=
  With[{b=Coefficient[v,x,1],d=Coefficient[u,x,0],e=Coefficient[u,x,1],f=Coefficient[w,x,0],g=Coefficient[w,x,1]},
  g*u^(m+1)*F^(c*v)/(b*c*e*Log[F]) /;
 EqQ[e*g*(m+1)-b*c*(e*f-d*g)*Log[F],0]] /;
FreeQ[{F,c,m},x] && LinearQ[{u,v,w},x]
```

4. $\int P_x\, u^m\, F^{c\,v}\, dx$ when $v == a + b\,x \land u == (d + e\,x)^n$

1: $\int P_x\, u^m\, F^{c\,v}\, dx$ when $v == a + b\,x \land u == (d + e\,x)^n \land m \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $v == a + b\,x \land u == (d + e\,x)^n \land m \in \mathbb{Z}$, then

$$\int P_x\, u^m\, F^{c\,v}\, dx \longrightarrow \int F^{c\,(a+b\,x)}\, \text{ExpandIntegrand}\left[P_x\, (d + e\,x)^{m\,n},\, x\right] dx$$

Program code:

```
Int[w_*u_^m_.*F_^(c_.*v_),x_Symbol] :=
  Int[ExpandIntegrand[w*NormalizePowerOfLinear[u,x]^m*F^(c*ExpandToSum[v,x]),x],x] /;
FreeQ[{F,c},x] && PolynomialQ[w,x] && LinearQ[v,x] && PowerOfLinearQ[u,x] && IntegerQ[m] && $UseGamma===True
```

```
Int[w_*u_^m_.*F_^(c_.*v_),x_Symbol] :=
  Int[ExpandIntegrand[F^(c*ExpandToSum[v,x]),w*NormalizePowerOfLinear[u,x]^m,x],x] /;
FreeQ[{F,c},x] && PolynomialQ[w,x] && LinearQ[v,x] && PowerOfLinearQ[u,x] && IntegerQ[m] && Not[$UseGamma===True]
```

2: $\int P_x\, u^m\, F^{c\,v}\, dx$ when $v == a + b\,x \land u == (d + e\,x)^n \land m \notin \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $v == a + b\,x \land u == (d + e\,x)^n \land m \notin \mathbb{Z}$, then

$$\int P_x\, u^m\, F^{c\,v}\, dx \longrightarrow \frac{\left((d + e\,x)^n\right)^m}{(d + e\,x)^{m\,n}} \int F^{c\,(a+b\,x)}\, \text{ExpandIntegrand}\left[P_x\,(d + e\,x)^{m\,n},\, x\right] dx$$

Program code:

```
Int[w_*u_^m_.*F_^(c_.*v_),x_Symbol] :=
  Module[{uu=NormalizePowerOfLinear[u,x],z},
  z=If[PowerQ[uu] && FreeQ[uu[[2]],x], uu[[1]]^(m*uu[[2]]), uu^m];
  uu^m/z*Int[ExpandIntegrand[w*z*F^(c*ExpandToSum[v,x]),x],x]] /;
FreeQ[{F,c,m},x] && PolynomialQ[w,x] && LinearQ[v,x] && PowerOfLinearQ[u,x] && Not[IntegerQ[m]]
```

5. $\int u\, F^{c\,(a+b\,x)}\, \text{Log}[d\,x]^n\, dx$

1: $\int F^{c\,(a+b\,x)}\, \text{Log}[d\,x]^n\, (e + h\,(f + g\,x)\, \text{Log}[d\,x])\, dx$ when $e == f\,h\,(n + 1) \land g\,h\,(n + 1) == b\,c\,e\,\text{Log}[F] \land n \neq -1$

Rule: If $e == f\,h\,(n + 1) \land g\,h\,(n + 1) == b\,c\,e\,\text{Log}[F] \land n \neq -1$, then

$$\int F^{c\,(a+b\,x)}\, \text{Log}[d\,x]^n\, (e + h\,(f + g\,x)\, \text{Log}[d\,x])\, dx \longrightarrow \frac{e\,x\,F^{c\,(a+b\,x)}\, \text{Log}[d\,x]^{n+1}}{n + 1}$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Log[d_.*x_]^n_.*(e_+h_.*(f_.+g_.*x_)*Log[d_.*x_]),x_Symbol] :=
  e*x*F^(c*(a+b*x))*Log[d*x]^(n+1)/(n+1) /;
FreeQ[{F,a,b,c,d,e,f,g,h,n},x] && EqQ[e-f*h*(n+1),0] && EqQ[g*h*(n+1)-b*c*e*Log[F],0] && NeQ[n,-1]
```

2: $\int x^m F^{c\,(a+b\,x)}\,\text{Log}[d\,x]^n\,(e+h\,(f+g\,x)\,\text{Log}[d\,x])\,dx$ when $e\,(m+1)=f\,h\,(n+1)\land g\,h\,(n+1)=b\,c\,e\,\text{Log}[F]\land n\neq-1$

Rule: If $e\,(m+1)=f\,h\,(n+1)\land g\,h\,(n+1)=b\,c\,e\,\text{Log}[F]\land n\neq-1$, then

$$\int x^m F^{c\,(a+b\,x)}\,\text{Log}[d\,x]^n\,(e+h\,(f+g\,x)\,\text{Log}[d\,x])\,dx\;\longrightarrow\;\frac{e\,x^{m+1}\,F^{c\,(a+b\,x)}\,\text{Log}[d\,x]^{n+1}}{n+1}$$

Program code:

```
Int[x_^m_.*F_^(c_.*(a_.+b_.*x_))*Log[d_.*x_]^n_.*(e_+h_.*(f_.+g_.*x_)*Log[d_.*x_]),x_Symbol] :=
  e*x^(m+1)*F^(c*(a+b*x))*Log[d*x]^(n+1)/(n+1) /;
FreeQ[{F,a,b,c,d,e,f,g,h,m,n},x] && EqQ[e*(m+1)-f*h*(n+1),0] && EqQ[g*h*(n+1)-b*c*e*Log[F],0] && NeQ[n,-1]
```

2. $\int u\, F^{a+b\,(c+d\,x)^n}\, dx$

1. $\int F^{a+b\,(c+d\,x)^n}\, dx$

1. $\int F^{a+b\,(c+d\,x)^n}\, dx$ when $\frac{2}{n} \in \mathbb{Z}$

1. $\int F^{a+b\,(c+d\,x)^n}\, dx$ when $\frac{2}{n} \in \mathbb{Z} \land n \in \mathbb{Z}$

1. $\int F^{a+b\,(c+d\,x)^n}\, dx$ when $\frac{2}{n} \in \mathbb{Z} \land n \in \mathbb{Z}^+$

1: $\int F^{a+b\,(c+d\,x)}\, dx$

Reference: G&R 2.311, CRC 519, A&S 4.2.54

Rule:

$$\int F^{a+b\,(c+d\,x)}\, dx \longrightarrow \frac{F^{a+b\,(c+d\,x)}}{b\, d\, \mathrm{Log}[F]}$$

Program code:

```
Int[F_^(a_.+b_.*(c_.+d_.*x_)),x_Symbol] :=
  F^(a+b*(c+d*x))/(b*d*Log[F]) /;
FreeQ[{F,a,b,c,d},x]
```

2. $\int F^{a+b\,(c+d\,x)^2}\, dx$

1: $\int F^{a+b\,(c+d\,x)^2}\, dx$ when $b > 0$

Basis: $\mathrm{Erfi}'[z] == \frac{2\, e^{z^2}}{\sqrt{\pi}}$

Rule: If $b > 0$, then

$$\int F^{a+b\,(c+d\,x)^2}\,dx\;\longrightarrow\;\frac{F^a\,\sqrt{\pi}\,\text{Erfi}\left[(c+d\,x)\,\sqrt{b\,\text{Log}[F]}\right]}{2\,d\,\sqrt{b\,\text{Log}[F]}}$$

Program code:

```
Int[F_^(a_.+b_.*(c_.+d_.*x_)^2),x_Symbol] :=
  F^a*Sqrt[Pi]*Erfi[(c+d*x)*Rt[b*Log[F],2]]/(2*d*Rt[b*Log[F],2]) /;
FreeQ[{F,a,b,c,d},x] && PosQ[b]
```

2: $\int F^{a+b\,(c+d\,x)^2}\,dx$ when $\neg\,(b>0)$

Basis: $\text{Erf}'[z] == \frac{2\,e^{-z^2}}{\sqrt{\pi}}$

Rule: If $\neg\,(b>0)$, then

$$\int F^{a+b\,(c+d\,x)^2}\,dx\;\longrightarrow\;\frac{F^a\,\sqrt{\pi}\,\text{Erf}\left[(c+d\,x)\,\sqrt{-b\,\text{Log}[F]}\right]}{2\,d\,\sqrt{-b\,\text{Log}[F]}}$$

Program code:

```
Int[F_^(a_.+b_.*(c_.+d_.*x_)^2),x_Symbol] :=
  F^a*Sqrt[Pi]*Erf[(c+d*x)*Rt[-b*Log[F],2]]/(2*d*Rt[-b*Log[F],2]) /;
FreeQ[{F,a,b,c,d},x] && NegQ[b]
```

2: $\int F^{a+b\,(c+d\,x)^n}\,dx$ when $\frac{2}{n}\in\mathbb{Z}\,\wedge\,n\in\mathbb{Z}^-$

Derivation: Integration by parts

Basis: $1 == \partial_x \frac{c+d\,x}{d}$

Rule: If $\frac{2}{n}\in\mathbb{Z}\,\wedge\,n\in\mathbb{Z}^-$, then

$$\int F^{a+b\,(c+d\,x)^n}\,dx \,\longrightarrow\, \frac{(c+d\,x)\,F^{a+b\,(c+d\,x)^n}}{d} - b\,n\,\text{Log}[F]\int (c+d\,x)^n\,F^{a+b\,(c+d\,x)^n}\,dx$$

Program code:

```
Int[F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  (c+d*x)*F^(a+b*(c+d*x)^n)/d -
  b*n*Log[F]*Int[(c+d*x)^n*F^(a+b*(c+d*x)^n),x] /;
FreeQ[{F,a,b,c,d},x] && IntegerQ[2/n] && ILtQ[n,0]
```

2: $\int F^{a+b\,(c+d\,x)^n}\,dx$ when $\frac{2}{n}\in\mathbb{Z}\land n\notin\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}^+$, then $F[(c+d\,x)^n] == \frac{k}{d}\left((c+d\,x)^{1/k}\right)^{k-1} F\left[\left((c+d\,x)^{1/k}\right)^{k\,n}\right]\partial_x (c+d\,x)^{1/k}$

Rule: If $\frac{2}{n}\in\mathbb{Z}\land n\notin\mathbb{Z}^+$, let $k=\text{Denominator}[n]$, then

$$\int F^{a+b\,x^n}\,dx \longrightarrow \frac{k}{d}\,\text{Subst}\left[\int x^{k-1}F^{a+b\,x^{k\,n}}\,dx,\ x,\ (c+d\,x)^{1/k}\right]$$

Program code:

```
Int[F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  With[{k=Denominator[n]},
  k/d*Subst[Int[x^(k-1)*F^(a+b*x^(k*n)),x],x,(c+d*x)^(1/k)]] /;
FreeQ[{F,a,b,c,d},x] && IntegerQ[2/n] && Not[IntegerQ[n]]
```

2: $\int F^{a+b\,(c+d\,x)^n}\,dx$ when $\frac{2}{n}\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c+d\,x)}{(-b\,(c+d\,x)^n\,\mathrm{Log}[F])^{1/n}} == 0$

Basis: $\partial_x \mathrm{Gamma}\left[\frac{1}{n},\,-b\,(c+d\,x)^n\,\mathrm{Log}[F]\right] == -\frac{d\,n\,F^{b\,(c+d\,x)^n}\,(-b\,(c+d\,x)^n\,\mathrm{Log}[F])^{\frac{1}{n}}}{c+d\,x}$

Rule: If $\frac{2}{n}\notin\mathbb{Z}$, then

$$\int F^{a+b\,(c+d\,x)^n}\,dx \longrightarrow -\frac{F^a\,(c+d\,x)\,\mathrm{Gamma}\left[\frac{1}{n},\,-b\,(c+d\,x)^n\,\mathrm{Log}[F]\right]}{d\,n\,(-b\,(c+d\,x)^n\,\mathrm{Log}[F])^{1/n}}$$

Program code:

```
Int[F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  -F^a*(c+d*x)*Gamma[1/n,-b*(c+d*x)^n*Log[F]]/(d*n*(-b*(c+d*x)^n*Log[F])^(1/n)) /;
FreeQ[{F,a,b,c,d,n},x] && Not[IntegerQ[2/n]]
```

2. $\int (e+f x)^m F^{a+b (c+d x)^n} \, dx$

   1. $\int (e+f x)^m F^{a+b (c+d x)^n} \, dx$ when $d e - c f == 0$

      1. $\int (e+f x)^m F^{a+b (c+d x)^n} \, dx$ when $d e - c f == 0 \land \frac{2 (m+1)}{n} \in \mathbb{Z}$

         1. $\int (c+d x)^m F^{a+b (c+d x)^n} \, dx$ when $\frac{2 (m+1)}{n} \in \mathbb{Z}$

            1: $\int (e+f x)^{n-1} F^{a+b (c+d x)^n} \, dx$ when $d e - c f == 0$

Derivation: Piecewise constant extraction and integration by substitution

Rule: If $d e - c f == 0$, then $\partial_x \frac{(e+f x)^n}{(c+d x)^n} == 0$

Basis: $(c+d x)^{n-1} F[(c+d x)^n] == \frac{1}{d n} F[(c+d x)^n] \partial_x (c+d x)^n$

Rule: If $d e - c f == 0$, then

$$\int (e+f x)^{n-1} F^{a+b (c+d x)^n} \, dx \longrightarrow \frac{(e+f x)^n F^{a+b (c+d x)^n}}{b f n (c+d x)^n \mathrm{Log}[F]}$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  (e+f*x)^n*F^(a+b*(c+d*x)^n)/(b*f*n*(c+d*x)^n*Log[F]) /;
FreeQ[{F,a,b,c,d,e,f,n},x] && EqQ[m,n-1] && EqQ[d*e-c*f,0]
```

2: $\int \frac{F^{a+b\,(c+d\,x)^n}}{e+f\,x}\,dx$ when $d\,e-c\,f==0$

Basis: $\text{ExpIntegralEi}'[z] == \frac{e^z}{z}$

Rule: If $d\,e-c\,f==0$, then

$$\int \frac{F^{a+b\,(c+d\,x)^n}}{e+f\,x}\,dx \longrightarrow \frac{F^a\,\text{ExpIntegralEi}\left[b\,(c+d\,x)^n\,\text{Log}[F]\right]}{f\,n}$$

Program code:

```
Int[F_^(a_.+b_.*(c_.+d_.*x_)^n_)/(e_.+f_.*x_),x_Symbol] :=
  F^a*ExpIntegralEi[b*(c+d*x)^n*Log[F]]/(f*n) /;
FreeQ[{F,a,b,c,d,e,f,n},x] && EqQ[d*e-c*f,0]
```

3: $\int (c+d\,x)^m\,F^{a+b\,(c+d\,x)^n}\,dx$ when $n==2\,(m+1)$

Derivation: Integration by substitution

Basis: If $n==2\,(m+1)$, then $(c+d\,x)^m\,F\left[(c+d\,x)^n\right] == \frac{1}{d\,(m+1)}\,F\left[\left((c+d\,x)^{m+1}\right)^2\right]\,\partial_x\,(c+d\,x)^{m+1}$

Rule: If $n==2\,(m+1)$, then

$$\int (c+d\,x)^m\,F^{a+b\,(c+d\,x)^n}\,dx \longrightarrow \frac{1}{d\,(m+1)}\,\text{Subst}\left[\int F^{a+b\,x^2}\,dx,\,x,\,(c+d\,x)^{m+1}\right]$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  1/(d*(m+1))*Subst[Int[F^(a+b*x^2),x],x,(c+d*x)^(m+1)] /;
FreeQ[{F,a,b,c,d,m,n},x] && EqQ[n,2*(m+1)]
```

4. $\int (c+d\,x)^m F^{a+b\,(c+d\,x)^n}\,dx$ when $\frac{2\,(m+1)}{n}\in\mathbb{Z}\wedge n\in\mathbb{Z}$

1: $\int (c+d\,x)^m F^{a+b\,(c+d\,x)^n}\,dx$ when $\frac{2\,(m+1)}{n}\in\mathbb{Z}\wedge n\in\mathbb{Z}\wedge(0<n<m+1\vee m<n<0)$

Reference: G&R 2.321.1, CRC 521, A&S 4.2.55

Derivation: Integration by parts

Basis: $(c+d\,x)^m F^{a+b\,(c+d\,x)^n} == (c+d\,x)^{m-n+1}\,\partial_x\frac{F^{a+b\,(c+d\,x)^n}}{b\,d\,n\,\mathrm{Log}[F]}$

Rule: If $\frac{2\,(m+1)}{n}\in\mathbb{Z}\wedge n\in\mathbb{Z}\wedge(0<n<m+1\vee m<n<0)$, then

$$\int (c+d\,x)^m F^{a+b\,(c+d\,x)^n}\,dx \longrightarrow \frac{(c+d\,x)^{m-n+1} F^{a+b\,(c+d\,x)^n}}{b\,d\,n\,\mathrm{Log}[F]}-\frac{m-n+1}{b\,n\,\mathrm{Log}[F]}\int (c+d\,x)^{m-n} F^{a+b\,(c+d\,x)^n}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  (c+d*x)^(m-n+1)*F^(a+b*(c+d*x)^n)/(b*d*n*Log[F]) -
  (m-n+1)/(b*n*Log[F])*Int[(c+d*x)^(m-n)*F^(a+b*(c+d*x)^n),x] /;
FreeQ[{F,a,b,c,d},x] && IntegerQ[2*(m+1)/n] && LtQ[0,(m+1)/n,5] && IntegerQ[n] && (LtQ[0,n,m+1] || LtQ[m,n,0])
```

```
Int[(c_.+d_.*x_)^m_.*F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  (c+d*x)^(m-n+1)*F^(a+b*(c+d*x)^n)/(b*d*n*Log[F]) -
  (m-n+1)/(b*n*Log[F])*Int[(c+d*x)^Simplify[m-n]*F^(a+b*(c+d*x)^n),x] /;
FreeQ[{F,a,b,c,d,m,n},x] && IntegerQ[2*Simplify[(m+1)/n]] && LtQ[0,Simplify[(m+1)/n],5] && Not[RationalQ[m]] && SumSimplerQ[m,-n]
```

2: $\int (c+d\,x)^m\,F^{a+b\,(c+d\,x)^n}\,dx$ when $\frac{2\,(m+1)}{n}\in\mathbb{Z}\wedge n\in\mathbb{Z}\wedge(n>0\wedge m<-1\vee 0<-n\leq m+1)$

Reference: G&R 2.324.1, CRC 523, A&S 4.2.56

Derivation: Integration by parts

Rule: If $\frac{2\,(m+1)}{n}\in\mathbb{Z}\wedge n\in\mathbb{Z}\wedge(n>0\wedge m<-1\vee 0<-n\leq m+1)$, then

$$\int (c+d\,x)^m\,F^{a+b\,(c+d\,x)^n}\,dx\,\longrightarrow\,\frac{(c+d\,x)^{m+1}\,F^{a+b\,(c+d\,x)^n}}{d\,(m+1)}-\frac{b\,n\,\text{Log}[F]}{m+1}\int (c+d\,x)^{m+n}\,F^{a+b\,(c+d\,x)^n}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  (c+d*x)^(m+1)*F^(a+b*(c+d*x)^n)/(d*(m+1)) -
  b*n*Log[F]/(m+1)*Int[(c+d*x)^(m+n)*F^(a+b*(c+d*x)^n),x] /;
FreeQ[{F,a,b,c,d},x] && IntegerQ[2*(m+1)/n] && LtQ[-4,(m+1)/n,5] && IntegerQ[n] && (GtQ[n,0] && LtQ[m,-1] || GtQ[-n,0] && LeQ[-n,m+1])
```

```
Int[(c_.+d_.*x_)^m_.*F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  (c+d*x)^(m+1)*F^(a+b*(c+d*x)^n)/(d*(m+1)) -
  b*n*Log[F]/(m+1)*Int[(c+d*x)^Simplify[m+n]*F^(a+b*(c+d*x)^n),x] /;
FreeQ[{F,a,b,c,d,m,n},x] && IntegerQ[2*Simplify[(m+1)/n]] && LtQ[-4,Simplify[(m+1)/n],5] && Not[RationalQ[m]] && SumSimplerQ[m,n]
```

5: $\int (c+d\,x)^m\, F^{a+b\,(c+d\,x)^n}\, dx$ when $\frac{2\,(m+1)}{n} \in \mathbb{Z}\ \land\ n \notin \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $(c+d\,x)^m\, F[(c+d\,x)^n] == \frac{k}{d}\, \left((c+d\,x)^{1/k}\right)^{k\,(m+1)-1}\, F\left[\left((c+d\,x)^{1/k}\right)^{k\,n}\right]\, \partial_x\, (c+d\,x)^{1/k}$

Rule: If $\frac{2\,(m+1)}{n} \in \mathbb{Z}\ \land\ n \notin \mathbb{Z}$, then

$$\int (c+d\,x)^m\, F^{a+b\,(c+d\,x)^n}\, dx \longrightarrow \frac{k}{d}\, \text{Subst}\left[\int x^{k\,(m+1)-1}\, F^{a+b\,x^{k\,n}}\, dx,\ x,\ (c+d\,x)^{1/k}\right]$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  With[{k=Denominator[n]},
  k/d*Subst[Int[x^(k*(m+1)-1)*F^(a+b*x^(k*n)),x],x,(c+d*x)^(1/k)]] /;
FreeQ[{F,a,b,c,d,m,n},x] && IntegerQ[2*(m+1)/n] && LtQ[0,(m+1)/n,5] && Not[IntegerQ[n]]
```

2: $\int (e+f x)^m F^{a+b (c+d x)^n} \, dx$ when $d e - c f == 0 \land \frac{2 (m+1)}{n} \in \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $d e - c f == 0$, then $\partial_x \frac{(e+f x)^m}{(c+d x)^m} == 0$

Rule: If $d e - c f == 0 \land \frac{2 (m+1)}{n} \in \mathbb{Z}$, then

$$\int (e+f x)^m F^{a+b (c+d x)^n} \, dx \longrightarrow \frac{(e+f x)^m}{(c+d x)^m} \int (c+d x)^m F^{a+b (c+d x)^n} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  (e+f*x)^m/(c+d*x)^m*Int[(c+d*x)^m*F^(a+b*(c+d*x)^n),x] /;
FreeQ[{F,a,b,c,d,e,f,m,n},x] && EqQ[d*e-c*f,0] && IntegerQ[2*Simplify[(m+1)/n]] && NeQ[f,d] && Not[IntegerQ[m]] && NeQ[c*e,0]
```

2: $\int (e+f x)^m F^{a+b (c+d x)^n} \, dx$ when $d e - c f == 0 \wedge \frac{2 (m+1)}{n} \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{c+d x}{(-b (c+d x)^n \text{Log}[F])^{1/n}} == 0$

Basis: $\partial_x \text{Gamma}\left[\frac{m+1}{n}, -b (c+d x)^n \text{Log}[F]\right] == -\frac{d n F^{b (c+d x)^n} (-b (c+d x)^n \text{Log}[F])^{\frac{m+1}{n}}}{c+d x}$

Note: This rule eliminates numerous steps and results in compact antiderivatives. When m or n is nonnumeric, *Mathematica* 8 and *Maple* 16 do not take advantage of it.

Note: To avoid introducing the incomplete gamma function when not absolutely necessary, apply the above substitution rule whenever $\frac{2 (m+1)}{n} \in \mathbb{Z}$.

Note: The special case $d e - c f == 0$ is important because $\partial_x \text{Gamma}[m, e+f x]$ equals $-f (e+f x)^{m-1} e^{-(e+f x)}$.

Rule: If $d e - c f == 0 \wedge \frac{2 (m+1)}{n} \notin \mathbb{Z}$, then

$$\int (e+f x)^m F^{a+b (c+d x)^n} \, dx \rightarrow -\frac{F^a (e+f x)^{m+1}}{f n} \text{ExpIntegralE}\left[1-\frac{m+1}{n}, -b (c+d x)^n \text{Log}[F]\right]$$

$$\int (e+f x)^m F^{a+b (c+d x)^n} \, dx \rightarrow -\frac{F^a (e+f x)^{m+1}}{f n (-b (c+d x)^n \text{Log}[F])^{\frac{m+1}{n}}} \text{Gamma}\left[\frac{m+1}{n}, -b (c+d x)^n \text{Log}[F]\right]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
(*-F^a*(e+f*x)^(m+1)/(f*n)*ExpIntegralE[1-(m+1)/n,-b*(c+d*x)^n*Log[F]] *)
  -F^a*(e+f*x)^(m+1)/(f*n*(-b*(c+d*x)^n*Log[F])^((m+1)/n))*Gamma[(m+1)/n,-b*(c+d*x)^n*Log[F]] /;
FreeQ[{F,a,b,c,d,e,f,m,n},x] && EqQ[d*e-c*f,0]
```

2. $\int (e+f x)^m F^{a+b (c+d x)^n} \, dx$ when $d e-c f \neq 0$

   1. $\int (e+f x)^m F^{a+b (c+d x)^2} \, dx$ when $d e-c f \neq 0$

      1: $\int (e+f x)^m F^{a+b (c+d x)^2} \, dx$ when $d e-c f \neq 0 \land m>1$

Derivation: Inverted integration by parts

Rule: If $d e-c f \neq 0 \land m>1$, then

$$\int (e+f x)^m F^{a+b (c+d x)^2} \, dx \longrightarrow$$

$$\frac{f (e+f x)^{m-1} F^{a+b (c+d x)^2}}{2 b d^2 \operatorname{Log}[F]}+\frac{d e-c f}{d} \int (e+f x)^{m-1} F^{a+b (c+d x)^2} \, dx-\frac{(m-1) f^2}{2 b d^2 \operatorname{Log}[F]} \int (e+f x)^{m-2} F^{a+b (c+d x)^2} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_*F_^(a_.+b_.*(c_.+d_.*x_)^2),x_Symbol] :=
  f*(e+f*x)^(m-1)*F^(a+b*(c+d*x)^2)/(2*b*d^2*Log[F]) +
  (d*e-c*f)/d*Int[(e+f*x)^(m-1)*F^(a+b*(c+d*x)^2),x] -
  (m-1)*f^2/(2*b*d^2*Log[F])*Int[(e+f*x)^(m-2)*F^(a+b*(c+d*x)^2),x] /;
FreeQ[{F,a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && FractionQ[m] && GtQ[m,1]
```

2: $\int (e+f\,x)^m\,F^{a+b\,(c+d\,x)^2}\,dx$ when $d\,e-c\,f\neq 0\,\wedge\,m<-1$

Derivation: Integration by parts

Rule: If $d\,e-c\,f\neq 0\,\wedge\,m<-1$, then

$$\int (e+f\,x)^m\,F^{a+b\,(c+d\,x)^2}\,dx\,\longrightarrow$$

$$\frac{f\,(e+f\,x)^{m+1}\,F^{a+b\,(c+d\,x)^2}}{(m+1)\,f^2}+\frac{2\,b\,d\,(d\,e-c\,f)\,\text{Log}[F]}{f^2\,(m+1)}\int (e+f\,x)^{m+1}\,F^{a+b\,(c+d\,x)^2}\,dx-\frac{2\,b\,d^2\,\text{Log}[F]}{f^2\,(m+1)}\int (e+f\,x)^{m+2}\,F^{a+b\,(c+d\,x)^2}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_*F_^(a_.+b_.*(c_.+d_.*x_)^2),x_Symbol] :=
  f*(e+f*x)^(m+1)*F^(a+b*(c+d*x)^2)/((m+1)*f^2) +
  2*b*d*(d*e-c*f)*Log[F]/(f^2*(m+1))*Int[(e+f*x)^(m+1)*F^(a+b*(c+d*x)^2),x] -
  2*b*d^2*Log[F]/(f^2*(m+1))*Int[(e+f*x)^(m+2)*F^(a+b*(c+d*x)^2),x] /;
FreeQ[{F,a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && LtQ[m,-1]
```

2: $\int (e+f x)^m F^{a+b (c+d x)^n} \, dx$ when $d e - c f \neq 0 \land n - 2 \in \mathbb{Z}^+ \land m < -1$

Derivation: Integration by parts

Basis: $(e+f x)^m == \partial_x \frac{(e+f x)^{m+1}}{f (m+1)}$

Rule: If $d e - c f \neq 0 \land n - 2 \in \mathbb{Z}^+ \land m < -1$, then

$$\int (e+f x)^m F^{a+b (c+d x)^n} \, dx \longrightarrow \frac{(e+f x)^{m+1} F^{a+b (c+d x)^n}}{f (m+1)} - \frac{b d n \operatorname{Log}[F]}{f (m+1)} \int (e+f x)^{m+1} (c+d x)^{n-1} F^{a+b (c+d x)^n} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_*F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  (e+f*x)^(m+1)*F^(a+b*(c+d*x)^n)/(f*(m+1)) -
  b*d*n*Log[F]/(f*(m+1))*Int[(e+f*x)^(m+1)*(c+d*x)^(n-1)*F^(a+b*(c+d*x)^n),x] /;
FreeQ[{F,a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && IGtQ[n,2] && LtQ[m,-1]
```

3. $\int (e+f x)^m F^{a+\frac{b}{c+d x}} \, dx$ when $d e-c f \neq 0 \wedge m \in \mathbb{Z}^-$

1: $\int \frac{F^{a+\frac{b}{c+d x}}}{e+f x} \, dx$ when $d e-c f \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{e+f x} = \frac{d}{f (c+d x)} - \frac{d e-c f}{f (c+d x) (e+f x)}$

Rule: If $d e-c f \neq 0$, then

$$\int \frac{F^{a+\frac{b}{c+d x}}}{e+f x} \, dx \longrightarrow \frac{d}{f} \int \frac{F^{a+\frac{b}{c+d x}}}{c+d x} \, dx - \frac{d e-c f}{f} \int \frac{F^{a+\frac{b}{c+d x}}}{(c+d x) (e+f x)} \, dx$$

Program code:

```
Int[F_^(a_.+b_./(c_.+d_.*x_))/(e_.+f_.*x_),x_Symbol] :=
  d/f*Int[F^(a+b/(c+d*x))/(c+d*x),x] -
  (d*e-c*f)/f*Int[F^(a+b/(c+d*x))/((c+d*x)*(e+f*x)),x] /;
FreeQ[{F,a,b,c,d,e,f},x] && NeQ[d*e-c*f,0]
```

2: $\int (e+f x)^m F^{a+\frac{b}{c+d x}} \, dx$ when $d e - c f \neq 0 \land m+1 \in \mathbb{Z}^-$

Derivation: Integration by parts

Basis: $(e+f x)^m == \partial_x \frac{(e+f x)^{m+1}}{f (m+1)}$

Note: Although resulting integrand appears more complicated than the original one, it is amenable to partial fraction expansion.

Rule: If $d e - c f \neq 0 \land m+1 \in \mathbb{Z}^-$, then

$$\int (e+f x)^m F^{a+\frac{b}{c+d x}} \, dx \longrightarrow \frac{(e+f x)^{m+1} F^{a+\frac{b}{c+d x}}}{f (m+1)} + \frac{b d \, \text{Log}[F]}{f (m+1)} \int \frac{(e+f x)^{m+1} F^{a+\frac{b}{c+d x}}}{(c+d x)^2} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_*F_^(a_.+b_./(c_.+d_.*x_)),x_Symbol] :=
  (e+f*x)^(m+1)*F^(a+b/(c+d*x))/(f*(m+1)) +
  b*d*Log[F]/(f*(m+1))*Int[(e+f*x)^(m+1)*F^(a+b/(c+d*x))/(c+d*x)^2,x] /;
FreeQ[{F,a,b,c,d,e,f},x] && NeQ[d*e-c*f,0] && ILtQ[m,-1]
```

X: $\int \frac{F^{a+b(c+d x)^n}}{e+f x} \, dx$ when $d e - c f \neq 0$

Rule: If $d e - c f \neq 0$, then

$$\int \frac{F^{a+b(c+d x)^n}}{e+f x} \, dx \longrightarrow \int \frac{F^{a+b(c+d x)^n}}{e+f x} \, dx$$

Program code:

```
Int[F_^(a_.+b_.*(c_.+d_.*x_)^n_)/(e_.+f_.*x_),x_Symbol] :=
  Unintegrable[F^(a+b*(c+d*x)^n)/(e+f*x),x] /;
FreeQ[{F,a,b,c,d,e,f,n},x] && NeQ[d*e-c*f,0]
```

3: $\int u^m F^v \, dx$ when $u == e + f x \wedge v == a + b x^n$

Derivation: Algebraic normalization

Rule: If $u == e + f x \wedge v == a + b x^n$, then

$$\int u^m F^v \, dx \longrightarrow \int (e + f x)^m F^{a+b x^n} \, dx$$

Program code:

```
Int[u_^m_.*F_^v_,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*F^ExpandToSum[v,x],x] /;
FreeQ[{F,m},x] && LinearQ[u,x] && BinomialQ[v,x] && Not[LinearMatchQ[u,x] && BinomialMatchQ[v,x]]
```

3. $\int P_x F^{a+b (c+d x)^n} dx$

1: $\int P_x F^{a+b (c+d x)^n} dx$

Derivation: Algebraic expansion

Rule:

$$\int P_x F^{a+b (c+d x)^n} dx \longrightarrow \int F^{a+b (c+d x)^n} \text{ExpandLinearProduct}[P_x, c, d, x] dx$$

Program code:

```
Int[u_*F_^(a_.+b_.*(c_.+d_.*x_)^n_),x_Symbol] :=
  Int[ExpandLinearProduct[F^(a+b*(c+d*x)^n),u,c,d,x],x] /;
FreeQ[{F,a,b,c,d,n},x] && PolynomialQ[u,x]
```

2: $\int P_x F^{a+b v} dx$ when $v == (c + d x)^n$

Derivation: Algebraic normalization

Rule: If $v == (c + d x)^n$, then

$$\int P_x F^{a+b v} dx \longrightarrow \int P_x F^{a+b (c+d x)^n} dx$$

Program code:

```
Int[u_.*F_^(a_.+b_.*v_),x_Symbol] :=
  Int[u*F^(a+b*NormalizePowerOfLinear[v,x]),x] /;
FreeQ[{F,a,b},x] && PolynomialQ[u,x] && PowerOfLinearQ[v,x] && Not[PowerOfLinearMatchQ[v,x]]
```

x: $\int P_x\, F^{a+b\, v^n}\, dx$ when $v == c + d\, x$

Derivation: Algebraic normalization

Rule: If $v == c + d\, x$, then

$$\int P_x\, F^{a+b\, v^n}\, dx \longrightarrow \int P_x\, F^{a+b\, (c+d\, x)^n}\, dx$$

Program code:

```
(* Int[u_.*F_^(a_.+b_.*v_^n_),x_Symbol] :=
  Int[u*F^(a+b*ExpandToSum[v,x]^n),x] /;
FreeQ[{F,a,b,n},x] && PolynomialQ[u,x] && LinearQ[v,x] && Not[LinearMatchQ[v,x]] *)
```

x: $\int P_x\, F^v\, dx$ when $v == a + b\, x^n$

Derivation: Algebraic normalization

Rule: If $v == a + b\, x^n$, then

$$\int P_x\, F^v\, dx \longrightarrow \int P_x\, F^{a+b\, x^n}\, dx$$

Program code:

```
(* Int[u_.*F_^u_,x_Symbol] :=
  Int[u*F^ExpandToSum[u,x],x] /;
FreeQ[F,x] && PolynomialQ[u,x] && BinomialQ[u,x] && Not[BinomialMatchQ[u,x]] *)
```

4: $\int \frac{F^{a+\frac{b}{c+d x}}}{(e+f x)(g+h x)} \, dx$ when $d e - c f == 0$

Derivation: Integration by substitution

Basis: If $d e - c f == 0$, then $\frac{F^{a+\frac{b}{c+d x}}}{(e+f x)(g+h x)} == -\frac{d}{f(d g-c h)} \frac{F^{a-\frac{b h}{d g-c h}+\frac{d b}{d g-c h} \frac{g+h x}{c+d x}}}{\frac{g+h x}{c+d x}} \partial_x \frac{g+h x}{c+d x}$

Rule: If $d e - c f == 0$, then

$$\int \frac{F^{a+\frac{b}{c+d x}}}{(e+f x)(g+h x)} \, dx \longrightarrow -\frac{d}{f(d g-c h)} \text{Subst}\left[\int \frac{F^{a-\frac{b h}{d g-c h}-\frac{-d b x}{d g-c h}}}{x} \, dx, x, \frac{g+h x}{c+d x}\right]$$

Program code:

```
Int[F_^(a_.+b_./(c_.+d_.*x_))/((e_.+f_.*x_)*(g_.+h_.*x_)),x_Symbol] :=
  -d/(f*(d*g-c*h))*Subst[Int[F^(a-b*h/(d*g-c*h)+d*b*x/(d*g-c*h))/x,x],x,(g+h*x)/(c+d*x)] /;
FreeQ[{F,a,b,c,d,e,f},x] && EqQ[d*e-c*f,0]
```

3. $\int u\, F^{e+f \frac{a+b x}{c+d x}}\, dx$

1. $\int (g+h x)^m\, F^{e+f \frac{a+b x}{c+d x}}\, dx$

1: $\int (g+h x)^m\, F^{e+f \frac{a+b x}{c+d x}}\, dx$ when $b c - a d == 0$

Derivation: Algebraic simplification

Basis: If $b c - a d == 0$, then $\frac{a+b x}{c+d x} == \frac{b}{d}$

Rule: If $b c - a d == 0$, then

$$\int (g+h x)^m\, F^{e+f \frac{a+b x}{c+d x}}\, dx \longrightarrow F^{e+f \frac{b}{d}} \int (g+h x)^m\, dx$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*F_^(e_.+f_.*(a_.+b_.*x_)/(c_.+d_.*x_)),x_Symbol] :=
  F^(e+f*b/d)*Int[(g+h*x)^m,x] /;
FreeQ[{F,a,b,c,d,e,f,g,h,m},x] && EqQ[b*c-a*d,0]
```

2. $\int (g+h\,x)^m F^{e+f\frac{a+b\,x}{c+d\,x}}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int (g+h\,x)^m F^{e+f\frac{a+b\,x}{c+d\,x}}\,dx$ when $b\,c-a\,d\neq 0\ \land\ d\,g-c\,h==0$

Derivation: Algebraic normalization

Basis: $e+f\,\frac{a+b\,x}{c+d\,x}==\frac{d\,e+b\,f}{d}-f\,\frac{b\,c-a\,d}{d\,(c+d\,x)}$

Rule: If $b\,c-a\,d\neq 0\ \land\ d\,g-c\,h==0$, then

$$\int (g+h\,x)^m F^{e+f\frac{a+b\,x}{c+d\,x}}\,dx\ \longrightarrow\ \int (g+h\,x)^m F^{\frac{d\,e+b\,f}{d}-f\frac{b\,c-a\,d}{d\,(c+d\,x)}}\,dx$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*F_^(e_.+f_.*(a_.+b_.*x_)/(c_.+d_.*x_)),x_Symbol] :=
  Int[(g+h*x)^m*F^((d*e+b*f)/d-f*(b*c-a*d)/(d*(c+d*x))),x] /;
FreeQ[{F,a,b,c,d,e,f,g,h,m},x] && NeQ[b*c-a*d,0] && EqQ[d*g-c*h,0]
```

2. $\int (g+h\,x)^m\, F^{e+f\,\frac{a+b\,x}{c+d\,x}}\, dx$ when $b\,c-a\,d\neq 0 \land d\,g-c\,h\neq 0$

1: $\int \frac{F^{e+f\,\frac{a+b\,x}{c+d\,x}}}{g+h\,x}\, dx$ when $b\,c-a\,d\neq 0 \land d\,g-c\,h\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{g+h\,x} == \frac{d}{h\,(c+d\,x)} - \frac{d\,g-c\,h}{h\,(c+d\,x)\,(g+h\,x)}$

Rule: If $b\,c-a\,d\neq 0 \land d\,g-c\,h\neq 0$, then

$$\int \frac{F^{e+f\,\frac{a+b\,x}{c+d\,x}}}{g+h\,x}\, dx \longrightarrow \frac{d}{h}\int \frac{F^{e+f\,\frac{a+b\,x}{c+d\,x}}}{c+d\,x}\, dx - \frac{d\,g-c\,h}{h}\int \frac{F^{e+f\,\frac{a+b\,x}{c+d\,x}}}{(c+d\,x)\,(g+h\,x)}\, dx$$

Program code:

```
Int[F_^(e_.+f_.*(a_.+b_.*x_)/(c_.+d_.*x_))/(g_.+h_.*x_),x_Symbol] :=
  d/h*Int[F^(e+f*(a+b*x)/(c+d*x))/(c+d*x),x] -
  (d*g-c*h)/h*Int[F^(e+f*(a+b*x)/(c+d*x))/((c+d*x)*(g+h*x)),x] /;
FreeQ[{F,a,b,c,d,e,f,g,h},x] && NeQ[b*c-a*d,0] && NeQ[d*g-c*h,0]
```

2: $\int (g+h x)^m F^{e+f \frac{a+b x}{c+d x}} \, dx$ when $b c-a d \neq 0 \land d g-c h \neq 0 \land m+1 \in \mathbb{Z}^-$

Derivation: Integration by parts

Basis: $(g+h x)^m == \partial_x \frac{(g+h x)^{m+1}}{h (m+1)}$

Note: Although resulting integrand appears more complicated than the original one, it is amenable to partial fraction expansion.

Rule: If $b c-a d \neq 0 \land d g-c h \neq 0 \land m+1 \in \mathbb{Z}^-$, then

$$\int (g+h x)^m F^{e+f \frac{a+b x}{c+d x}} \, dx \longrightarrow \frac{(g+h x)^{m+1} F^{e+f \frac{a+b x}{c+d x}}}{h (m+1)} - \frac{f (b c-a d) \operatorname{Log}[F]}{h (m+1)} \int \frac{(g+h x)^{m+1} F^{e+f \frac{a+b x}{c+d x}}}{(c+d x)^2} \, dx$$

Program code:

```
Int[(g_.+h_.*x_)^m_*F_^(e_.+f_.*(a_.+b_.*x_)/(c_.+d_.*x_)),x_Symbol] :=
  (g+h*x)^(m+1)*F^(e+f*(a+b*x)/(c+d*x))/(h*(m+1)) -
  f*(b*c-a*d)*Log[F]/(h*(m+1))*Int[(g+h*x)^(m+1)*F^(e+f*(a+b*x)/(c+d*x))/(c+d*x)^2,x] /;
FreeQ[{F,a,b,c,d,e,f,g,h},x] && NeQ[b*c-a*d,0] && NeQ[d*g-c*h,0] && ILtQ[m,-1]
```

2: $\int \frac{F^{e+f \frac{a+b x}{c+d x}}}{(g+h x) (i+j x)} \, dx$ when $d g-c h == 0$

Derivation: Integration by substitution

–

Basis: If $d g-c h == 0$, then $\frac{F^{e+f \frac{a+b x}{c+d x}}}{(g+h x) (i+j x)} == -\frac{d}{h (d i-c j)} \frac{F^{e+\frac{f (b i-a j)}{d i-c j}-\frac{(b c-a d) f}{d i-c j} \frac{i+j x}{c+d x}}}{\frac{i+j x}{c+d x}} \partial_x \frac{i+j x}{c+d x}$

Rule: If $d g-c h == 0$, then

$$\int \frac{F^{e+f\frac{a+b x}{c+d x}}}{(g+h x)(i+j x)}\,dx \longrightarrow -\frac{d}{h (d i-c j)}\,\text{Subst}\left[\int \frac{F^{e+\frac{f (b i-a j)}{d i-c j}-\frac{(b c-a d) f x}{d i-c j}}}{x}\,dx,\,x,\,\frac{i+j x}{c+d x}\right]$$

Program code:

```
Int[F_^(e_.+f_.*(a_.+b_.*x_)/(c_.+d_.*x_))/((g_.+h_.*x_)*(i_.+j_.*x_)),x_Symbol] :=
  -d/(h*(d*i-c*j))*Subst[Int[F^(e+f*(b*i-a*j)/(d*i-c*j)-(b*c-a*d)*f*x/(d*i-c*j))/x,x],x,(i+j*x)/(c+d*x)] /;
FreeQ[{F,a,b,c,d,e,f,g,h},x] && EqQ[d*g-c*h,0]
```

4. $\int u\,F^{a+b x+c x^2}\,dx$

1. $\int F^{a+b x+c x^2}\,dx$

1: $\int F^{a+b x+c x^2}\,dx$

Derivation: Algebraic expansion

Basis: $a+b x+c x^2 == \frac{4 a c-b^2}{4 c}+\frac{(b+2 c x)^2}{4 c}$

Basis: $F^{z+w} == F^z F^w$

Rule:

$$\int F^{a+b x+c x^2}\,dx \longrightarrow F^{\frac{4 a c-b^2}{4 c}} \int F^{\frac{(b+2 c x)^2}{4 c}}\,dx$$

Program code:

```
Int[F_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  F^(a-b^2/(4*c))*Int[F^((b+2*c*x)^2/(4*c)),x] /;
FreeQ[{F,a,b,c},x]
```

2: $\int F^v \, dx$ when $v == a + b x + c x^2$

Derivation: Algebraic normalization

Rule: If $v == a + b x + c x^2$, then

$$\int F^v \, dx \longrightarrow \int F^{a+b x+c x^2} \, dx$$

Program code:

```
Int[F_^v_,x_Symbol] :=
  Int[F^ExpandToSum[v,x],x] /;
FreeQ[F,x] && QuadraticQ[v,x] && Not[QuadraticMatchQ[v,x]]
```

2. $\int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx$

   1. $\int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx$ when $b\,e-2\,c\,d==0$

      1. $\int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx$ when $b\,e-2\,c\,d==0\land m>0$

         1: $\int (d+e\,x)\,F^{a+b\,x+c\,x^2}\,dx$ when $b\,e-2\,c\,d==0$

Derivation: Integration by substitution

Rule: If $b\,e-2\,c\,d==0$, then

$$\int (d+e\,x)\,F^{a+b\,x+c\,x^2}\,dx\;\longrightarrow\;\frac{e\,F^{a+b\,x+c\,x^2}}{2\,c\,\mathrm{Log}[F]}$$

Program code:

```
Int[(d_.+e_.*x_)*F_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*F^(a+b*x+c*x^2)/(2*c*Log[F]) /;
FreeQ[{F,a,b,c,d,e},x] && EqQ[b*e-2*c*d,0]
```

2: $\int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx$ when $b\,e-2\,c\,d=0\,\wedge\,m>1$

Derivation: Inverted integration by parts

Rule: If $b\,e-2\,c\,d=0\,\wedge\,m>1$, then

$$\int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx\,\longrightarrow\,\frac{e\,(d+e\,x)^{m-1}\,F^{a+b\,x+c\,x^2}}{2\,c\,\mathrm{Log}[F]}-\frac{(m-1)\,e^2}{2\,c\,\mathrm{Log}[F]}\int (d+e\,x)^{m-2}\,F^{a+b\,x+c\,x^2}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*F_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*(d+e*x)^(m-1)*F^(a+b*x+c*x^2)/(2*c*Log[F]) -
  (m-1)*e^2/(2*c*Log[F])*Int[(d+e*x)^(m-2)*F^(a+b*x+c*x^2),x] /;
FreeQ[{F,a,b,c,d,e},x] && EqQ[b*e-2*c*d,0] && GtQ[m,1]
```

2. $\int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx$ when $b\,e-2\,c\,d=0\,\wedge\,m<0$

1: $\int \frac{F^{a+b\,x+c\,x^2}}{d+e\,x}\,dx$ when $b\,e-2\,c\,d=0$

Rule: If $b\,e-2\,c\,d=0$, then

$$\int \frac{F^{a+b\,x+c\,x^2}}{d+e\,x}\,dx\,\longrightarrow\,\frac{1}{2\,e}\,F^{a-\frac{b^2}{4\,c}}\,\mathrm{ExpIntegralEi}\left[\frac{(b+2\,c\,x)^2\,\mathrm{Log}[F]}{4\,c}\right]$$

Program code:

```
Int[F_^(a_.+b_.*x_+c_.*x_^2)/(d_.+e_.*x_),x_Symbol] :=
  1/(2*e)*F^(a-b^2/(4*c))*ExpIntegralEi[(b+2*c*x)^2*Log[F]/(4*c)] /;
FreeQ[{F,a,b,c,d,e},x] && EqQ[b*e-2*c*d,0]
```

2: $\int (d+e\,x)^m F^{a+b\,x+c\,x^2}\,dx$ when $b\,e-2\,c\,d==0\land m<-1$

Derivation: Integration by parts

Rule: If $b\,e-2\,c\,d==0\land m<-1$, then

$$\int (d+e\,x)^m F^{a+b\,x+c\,x^2}\,dx\longrightarrow\frac{(d+e\,x)^{m+1} F^{a+b\,x+c\,x^2}}{e\,(m+1)}-\frac{2\,c\,\text{Log}[F]}{e^2\,(m+1)}\int (d+e\,x)^{m+2} F^{a+b\,x+c\,x^2}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*F_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  (d+e*x)^(m+1)*F^(a+b*x+c*x^2)/(e*(m+1)) -
  2*c*Log[F]/(e^2*(m+1))*Int[(d+e*x)^(m+2)*F^(a+b*x+c*x^2),x] /;
FreeQ[{F,a,b,c,d,e},x] && EqQ[b*e-2*c*d,0] && LtQ[m,-1]
```

2. $\int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx$ when $b\,e-2\,c\,d\neq 0$

   1. $\int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx$ when $b\,e-2\,c\,d\neq 0\,\land\,m>0$

      1: $\int (d+e\,x)\,F^{a+b\,x+c\,x^2}\,dx$ when $b\,e-2\,c\,d\neq 0$

Derivation: Inverted integration by parts

Rule: If $b\,e-2\,c\,d\neq 0$, then

$$\int (d+e\,x)\,F^{a+b\,x+c\,x^2}\,dx\;\longrightarrow\;\frac{e\,F^{a+b\,x+c\,x^2}}{2\,c\,\mathrm{Log}[F]}-\frac{b\,e-2\,c\,d}{2\,c}\int F^{a+b\,x+c\,x^2}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)*F_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*F^(a+b*x+c*x^2)/(2*c*Log[F]) -
  (b*e-2*c*d)/(2*c)*Int[F^(a+b*x+c*x^2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[b*e-2*c*d,0]
```

2: $\int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx$ when $b\,e-2\,c\,d\neq 0\,\wedge\,m>1$

Derivation: Inverted integration by parts

Rule: If $b\,e-2\,c\,d\neq 0\,\wedge\,m>1$, then

$$\int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx\,\longrightarrow$$

$$\frac{e\,(d+e\,x)^{m-1}\,F^{a+b\,x+c\,x^2}}{2\,c\,\mathrm{Log}[F]}-\frac{b\,e-2\,c\,d}{2\,c}\int (d+e\,x)^{m-1}\,F^{a+b\,x+c\,x^2}\,dx-\frac{(m-1)\,e^2}{2\,c\,\mathrm{Log}[F]}\int (d+e\,x)^{m-2}\,F^{a+b\,x+c\,x^2}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*F_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  e*(d+e*x)^(m-1)*F^(a+b*x+c*x^2)/(2*c*Log[F]) -
  (b*e-2*c*d)/(2*c)*Int[(d+e*x)^(m-1)*F^(a+b*x+c*x^2),x] -
  (m-1)*e^2/(2*c*Log[F])*Int[(d+e*x)^(m-2)*F^(a+b*x+c*x^2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[b*e-2*c*d,0] && GtQ[m,1]
```

2: $\int (d+e\,x)^m\, F^{a+b\,x+c\,x^2}\, dx$ when $b\,e-2\,c\,d\neq 0\,\land\,m<-1$

Derivation: Integration by parts

Rule: If $b\,e-2\,c\,d\neq 0\,\land\,m<-1$, then

$$\int (d+e\,x)^m\, F^{a+b\,x+c\,x^2}\, dx \longrightarrow$$
$$\frac{(d+e\,x)^{m+1}\, F^{a+b\,x+c\,x^2}}{e\,(m+1)}-\frac{(b\,e-2\,c\,d)\,\mathrm{Log}[F]}{e^2\,(m+1)}\int (d+e\,x)^{m+1}\, F^{a+b\,x+c\,x^2}\, dx-\frac{2\,c\,\mathrm{Log}[F]}{e^2\,(m+1)}\int (d+e\,x)^{m+2}\, F^{a+b\,x+c\,x^2}\, dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_*F_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  (d+e*x)^(m+1)*F^(a+b*x+c*x^2)/(e*(m+1)) -
  (b*e-2*c*d)*Log[F]/(e^2*(m+1))*Int[(d+e*x)^(m+1)*F^(a+b*x+c*x^2),x] -
  2*c*Log[F]/(e^2*(m+1))*Int[(d+e*x)^(m+2)*F^(a+b*x+c*x^2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[b*e-2*c*d,0] && LtQ[m,-1]
```

X: $\int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx$

Derivation: Algebraic normalization

Rule: If $u == d+e\,x \land v == a+b\,x+c\,x^2$, then

$$\int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx \longrightarrow \int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*F_^(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  Unintegrable[(d+e*x)^m*F^(a+b*x+c*x^2),x] /;
FreeQ[{F,a,b,c,d,e,m},x]
```

4: $\int u^m\,F^v\,dx$ when $u == d+e\,x \land v == a+b\,x+c\,x^2$

Derivation: Algebraic normalization

Rule: If $u == d+e\,x \land v == a+b\,x+c\,x^2$, then

$$\int u^m\,F^v\,dx \longrightarrow \int (d+e\,x)^m\,F^{a+b\,x+c\,x^2}\,dx$$

Program code:

```
Int[u_^m_.*F_^v_,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*F^ExpandToSum[v,x],x] /;
FreeQ[{F,m},x] && LinearQ[u,x] && QuadraticQ[v,x] && Not[LinearMatchQ[u,x] && QuadraticMatchQ[v,x]]
```

5. $\int u \left(a + b \left(F^{e\,(c+d\,x)}\right)^n\right)^p\, dx$

1: $\int x^m F^{e\,(c+d\,x)} \left(a + b\, F^{2\,e\,(c+d\,x)}\right)^p\, dx$ when $m > 0 \land p \in \mathbb{Z}^-$

Derivation: Integration by parts

Rule: If $m > 0 \land p \in \mathbb{Z}^-$, then

$$\int x^m F^{e\,(c+d\,x)} \left(a + b\, F^{2\,e\,(c+d\,x)}\right)^p dx \longrightarrow x^m \int F^{e\,(c+d\,x)} \left(a + b\, F^{2\,e\,(c+d\,x)}\right)^p dx - m \int x^{m-1} \left(\int F^{e\,(c+d\,x)} \left(a + b\, F^{2\,e\,(c+d\,x)}\right)^p dx\right) dx$$

Program code:

```
Int[x_^m_.*F_^(e_.*(c_.+d_.*x_))*(a_.+b_.*F_^v_)^p_,x_Symbol] :=
  With[{u=IntHide[F^(e*(c+d*x))*(a+b*F^v)^p,x]},
  Dist[x^m,u,x] - m*Int[x^(m-1)*u,x]] /;
FreeQ[{F,a,b,c,d,e},x] && EqQ[v,2*e*(c+d*x)] && GtQ[m,0] && ILtQ[p,0]
```

2. $\int \left(G^{h\,(f+g\,x)}\right)^m \left(a+b\left(F^{e\,(c+d\,x)}\right)^n\right)^p\,dx$ when $d\,e\,n\,\text{Log}[F] == g\,h\,m\,\text{Log}[G]$

1: $\int \left(F^{e\,(c+d\,x)}\right)^n \left(a+b\left(F^{e\,(c+d\,x)}\right)^n\right)^p\,dx$

Derivation: Integration by substitution

Basis: $\left(F^{e\,(c+d\,x)}\right)^n \left(a+b\left(F^{e\,(c+d\,x)}\right)^n\right)^p == \frac{1}{d\,e\,n\,\text{Log}[F]}\,\text{Subst}\left[(a+b\,x)^p,\,x,\,\left(F^{e\,(c+d\,x)}\right)^n\right]\,\partial_x\left(F^{e\,(c+d\,x)}\right)^n$

Rule:

$$\int \left(F^{e\,(c+d\,x)}\right)^n \left(a+b\left(F^{e\,(c+d\,x)}\right)^n\right)^p\,dx \longrightarrow \frac{1}{d\,e\,n\,\text{Log}[F]}\,\text{Subst}\left[\int (a+b\,x)^p\,dx,\,x,\,\left(F^{e\,(c+d\,x)}\right)^n\right]$$

Program code:

```
Int[(F_^(e_.*(c_.+d_.*x_)))^n_.*(a_+b_.*(F_^(e_.*(c_.+d_.*x_)))^n_.)^p_.,x_Symbol] :=
  1/(d*e*n*Log[F])*Subst[Int[(a+b*x)^p,x],x,(F^(e*(c+d*x)))^n] /;
FreeQ[{F,a,b,c,d,e,n,p},x]
```

2: $\int \left(G^{h\,(f+g\,x)}\right)^m \left(a+b\left(F^{e\,(c+d\,x)}\right)^n\right)^p\,dx$ when $d\,e\,n\,\text{Log}[F]==g\,h\,m\,\text{Log}[G]$

Derivation: Piecewise constant extraction

Basis: If $d\,e\,n\,\text{Log}[F]==g\,h\,m\,\text{Log}[G]$, then $\partial_x \frac{\left(G^{h\,(f+g\,x)}\right)^m}{\left(F^{e\,(c+d\,x)}\right)^n}==0$

Rule: If $d\,e\,n\,\text{Log}[F]==g\,h\,m\,\text{Log}[G]$, then

$$\int \left(G^{h\,(f+g\,x)}\right)^m \left(a+b\left(F^{e\,(c+d\,x)}\right)^n\right)^p dx \to \frac{\left(G^{h\,(f+g\,x)}\right)^m}{\left(F^{e\,(c+d\,x)}\right)^n}\int \left(F^{e\,(c+d\,x)}\right)^n \left(a+b\left(F^{e\,(c+d\,x)}\right)^n\right)^p dx$$

Program code:

```
Int[(G_^(h_.*(f_.+g_.*x_)))^m_.*(a_+b_.*(F_^(e_.*(c_.+d_.*x_)))^n_.)^p_.,x_Symbol] :=
  (G^(h*(f+g*x)))^m/(F^(e*(c+d*x)))^n*Int[(F^(e*(c+d*x)))^n*(a+b*(F^(e*(c+d*x)))^n)^p,x] /;
FreeQ[{F,G,a,b,c,d,e,f,g,h,m,n,p},x] && EqQ[d*e*n*Log[F],g*h*m*Log[G]]
```

3. $\int G^{h\,(f+g\,x)} \left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$

1. $\int G^{h\,(f+g\,x)} \left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $\frac{g\,h\,\text{Log}[G]}{d\,e\,\text{Log}[F]}\in\mathbb{R}$

1: $\int G^{h\,(f+g\,x)} \left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $\text{Abs}\left[\frac{g\,h\,\text{Log}[G]}{d\,e\,\text{Log}[F]}\right]\geq 1$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}\ \land\ k\,\frac{g\,h\,\text{Log}[G]}{d\,e\,\text{Log}[F]}\in\mathbb{Z}$, then

$$G^{h\,(f+g\,x)} \left(a+b\,F^{e\,(c+d\,x)}\right)^p==\frac{k\,G^{f\,h-\frac{c\,g\,h}{d}}}{d\,e\,\text{Log}[F]}\,\text{Subst}\left[x^{k\,\frac{g\,h\,\text{Log}[G]}{d\,e\,\text{Log}[F]}-1}\left(a+b\,x^k\right)^p,\ x,\ F^{\frac{e\,(c+d\,x)}{k}}\right]\partial_x F^{\frac{e\,(c+d\,x)}{k}}$$

Rule: If $\text{Abs}\left[\frac{g\,h\,\text{Log}[G]}{d\,e\,\text{Log}[F]}\right]\geq 1$, then

$$\int G^{h\,(f+g\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx \to \frac{k\,G^{f\,h-\frac{c\,g\,h}{d}}}{d\,e\,\text{Log}[F]}\,\text{Subst}\left[\int x^{k\,\frac{g\,h\,\text{Log}[G]}{d\,e\,\text{Log}[F]}-1}\,\left(a+b\,x^k\right)^p\,dx,\,x,\,F^{\frac{e\,(c+d\,x)}{k}}\right]$$

Program code:

```
Int[G_^(h_.(f_.+g_.*x_))*(a_+b_.*F_^(e_.*(c_.+d_.*x_)))^p_.,x_Symbol] :=
  With[{m=FullSimplify[g*h*Log[G]/(d*e*Log[F])]},
  Denominator[m]*G^(f*h-c*g*h/d)/(d*e*Log[F])*Subst[Int[x^(Numerator[m]-1)*(a+b*x^Denominator[m])^p,x],x,F^(e*(c+d*x)/Denominator[m])]
 LeQ[m,-1] || GeQ[m,1]] /;
FreeQ[{F,G,a,b,c,d,e,f,g,h,p},x]
```

2: $\int G^{h\,(f+g\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $\text{Abs}\left[\frac{d\,e\,\text{Log}[F]}{g\,h\,\text{Log}[G]}\right]>1$

Derivation: Integration by substitution

Basis: If $k\in\mathbb{Z}\,\wedge\,k\,\frac{d\,e\,\text{Log}[F]}{g\,h\,\text{Log}[G]}\in\mathbb{Z}$, then

$$G^{h\,(f+g\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p == \frac{k}{g\,h\,\text{Log}[G]}\,\text{Subst}\left[x^{k-1}\,\left(a+b\,F^{c\,e-\frac{d\,e\,f}{g}}\,x^{k\,\frac{d\,e\,\text{Log}[F]}{g\,h\,\text{Log}[G]}}\right)^p,\,x,\,G^{\frac{h\,(f+g\,x)}{k}}\right]\,\partial_x\,G^{\frac{h\,(f+g\,x)}{k}}$$

Rule: If $\text{Abs}\left[\frac{d\,e\,\text{Log}[F]}{g\,h\,\text{Log}[G]}\right]>1$, then

$$\int G^{h\,(f+g\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx \to \frac{k}{g\,h\,\text{Log}[G]}\,\text{Subst}\left[\int x^{k-1}\,\left(a+b\,F^{c\,e-\frac{d\,e\,f}{g}}\,x^{k\,\frac{d\,e\,\text{Log}[F]}{g\,h\,\text{Log}[G]}}\right)^p\,dx,\,x,\,G^{\frac{h\,(f+g\,x)}{k}}\right]$$

Program code:

```
Int[G_^(h_.(f_.+g_.*x_))*(a_+b_.*F_^(e_.*(c_.+d_.*x_)))^p_.,x_Symbol] :=
  With[{m=FullSimplify[d*e*Log[F]/(g*h*Log[G])]},
  Denominator[m]/(g*h*Log[G])*Subst[Int[x^(Denominator[m]-1)*(a+b*F^(c*e-d*e*f/g)*x^Numerator[m])^p,x],x,G^(h*(f+g*x)/Denominator[m])]
 LtQ[m,-1] || GtQ[m,1]] /;
FreeQ[{F,G,a,b,c,d,e,f,g,h,p},x]
```

2. $\int G^{h\,(f+g\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $\frac{g\,h\,Log[G]}{d\,e\,Log[F]}\notin\mathbb{R}$

1: $\int G^{h\,(f+g\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $p\in\mathbb{Z}^+$

Rule: If $p\in\mathbb{Z}^+$, then

$$\int G^{h\,(f+g\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx\;\longrightarrow\;\int Expand\left[G^{h\,(f+g\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\right]dx$$

Program code:

```
Int[G_^(h_.(f_.+g_.*x_))*(a_+b_.*F_^(e_.*(c_.+d_.*x_)))^p_.,x_Symbol] :=
  Int[Expand[G^(h*(f+g*x))*(a+b*F^(e*(c+d*x)))^p,x],x] /;
FreeQ[{F,G,a,b,c,d,e,f,g,h},x] && IGtQ[p,0]
```

2: $\int G^{h\,(f+g\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $p\in\mathbb{Z}^-\lor a>0$

Rule: If $p\in\mathbb{Z}^-\lor a>0$, then

$$\int G^{h\,(f+g\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx\;\longrightarrow\;\frac{a^p\,G^{h\,(f+g\,x)}}{g\,h\,Log[G]}\,Hypergeometric2F1\left[-p,\,\frac{g\,h\,Log[G]}{d\,e\,Log[F]},\,\frac{g\,h\,Log[G]}{d\,e\,Log[F]}+1,\,-\frac{b}{a}\,F^{e\,(c+d\,x)}\right]$$

Program code:

```
Int[G_^(h_.(f_.+g_.*x_))*(a_+b_.*F_^(e_.*(c_.+d_.*x_)))^p_,x_Symbol] :=
  a^p*G^(h*(f+g*x))/(g*h*Log[G])*Hypergeometric2F1[-p,g*h*Log[G]/(d*e*Log[F]),g*h*Log[G]/(d*e*Log[F])+1,Simplify[-b/a*F^(e*(c+d*x))]]
FreeQ[{F,G,a,b,c,d,e,f,g,h,p},x] && (ILtQ[p,0] || GtQ[a,0])
```

3: $\int G^{h (f+g x)} \left(a + b F^{e (c+d x)}\right)^p dx$ when $\neg (p \in \mathbb{Z}^- \lor a > 0)$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\left(a+b F^{e (c+d x)}\right)^p}{\left(1+\frac{b F^{e (c+d x)}}{a}\right)^p} == 0$

Rule: If $\neg (p \in \mathbb{Z}^- \lor a > 0)$, then

$$\int G^{h (f+g x)} \left(a + b F^{e (c+d x)}\right)^p dx \longrightarrow \frac{\left(a + b F^{e (c+d x)}\right)^p}{\left(1 + \frac{b}{a} F^{e (c+d x)}\right)^p} \int G^{h (f+g x)} \left(1 + \frac{b}{a} F^{e (c+d x)}\right)^p dx$$

Program code:

```
Int[G_^(h_.(f_.+g_.*x_))*(a_+b_.*F_^(e_.*(c_.+d_.*x_)))^p_,x_Symbol] :=
  (a+b*F^(e*(c+d*x)))^p/(1+(b/a)*F^(e*(c+d*x)))^p*Int[G^(h*(f+g*x))*(1+b/a*F^(e*(c+d*x)))^p,x] /;
FreeQ[{F,G,a,b,c,d,e,f,g,h,p},x] && Not[ILtQ[p,0] || GtQ[a,0]]
```

3: $\int G^{h u} \left(a + b F^{e v}\right)^p dx$ when $u == f + g x \land v == c + d x$

Derivation: Algebraic normalization

Rule: If $u == f + g x \land v == c + d x$, then

$$\int G^{h u} \left(a + b F^{e v}\right)^p dx \longrightarrow \int G^{h (f+g x)} \left(a + b F^{e (c+d x)}\right)^p dx$$

Program code:

```
Int[G_^(h_.u_)*(a_+b_.*F_^(e_.*v_))^p_,x_Symbol] :=
  Int[G^(h*ExpandToSum[u,x])*(a+b*F^(e*ExpandToSum[v,x]))^p,x] /;
FreeQ[{F,G,a,b,e,h,p},x] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```

4. $\int (e+f x)^m \left(a+b F^{g (i+j x)}\right)^p \left(c+d F^{h (i+j x)}\right)^q dx$ when $(p \mid q) \in \mathbb{Z} \land \frac{g}{h} \in \mathbb{R}$

x: $\int \frac{(c+d x)^m F^{g (e+f x)}}{a+b F^{h (e+f x)}} dx$ when $0 \leq \frac{g}{h} - 1 < \frac{g}{h}$

Derivation: Algebraic expansion

Basis: $\frac{F^{g z}}{a+b F^{h z}} == \frac{F^{(g-h) z}}{b} - \frac{a F^{(g-h) z}}{b \left(a+b F^{h z}\right)}$

Rule: If $0 \leq \frac{g}{h} - 1 < \frac{g}{h}$, then

$$\int \frac{(c+d x)^m F^{g (e+f x)}}{a+b F^{h (e+f x)}} dx \longrightarrow \frac{1}{b} \int (c+d x)^m F^{(g-h) (e+f x)} dx - \frac{a}{b} \int \frac{(c+d x)^m F^{(g-h) (e+f x)}}{a+b F^{h (e+f x)}} dx$$

Program code:

```
(* Int[(c_.+d_.*x_)^m_.*F_^(g_.*(e_.+f_.*x_))/(a_+b_.*F_^(h_.*(e_.+f_.*x_))),x_Symbol] :=
  1/b*Int[(c+d*x)^m*F^((g-h)*(e+f*x)),x] -
  a/b*Int[(c+d*x)^m*F^((g-h)*(e+f*x))/(a+b*F^(h*(e+f*x))),x] /;
FreeQ[{F,a,b,c,d,e,f,g,h,m},x] && LeQ[0,g/h-1,g/h] *)
```

x: $\int \frac{(c+d\,x)^m\,F^{g\,(e+f\,x)}}{a+b\,F^{h\,(e+f\,x)}}\,dx$ when $\frac{g}{h}<\frac{g}{h}+1\leq 0$

Derivation: Algebraic expansion

Basis: $\frac{F^{g\,z}}{a+b\,F^{h\,z}} == \frac{F^{g\,z}}{a} - \frac{b\,F^{(g+h)\,z}}{a\,(a+b\,F^{h\,z})}$

Rule: If $\frac{g}{h}<\frac{g}{h}+1\leq 0$, then

$$\int \frac{(c+d\,x)^m\,F^{g\,(e+f\,x)}}{a+b\,F^{h\,(e+f\,x)}}\,dx \longrightarrow \frac{1}{a}\int (c+d\,x)^m\,F^{g\,(e+f\,x)}\,dx - \frac{b}{a}\int \frac{(c+d\,x)^m\,F^{(g+h)\,(e+f\,x)}}{a+b\,F^{h\,(e+f\,x)}}\,dx$$

Program code:

```
(* Int[(c_.+d_.*x_)^m_.*F_^(g_.*(e_.+f_.*x_))/(a_+b_.*F_^(h_.*(e_.+f_.*x_))),x_Symbol] :=
  1/a*Int[(c+d*x)^m*F^(g*(e+f*x)),x] -
  b/a*Int[(c+d*x)^m*F^((g+h)*(e+f*x))/(a+b*F^(h*(e+f*x))),x] /;
FreeQ[{F,a,b,c,d,e,f,g,h,m},x] && LeQ[g/h,g/h+1,0] *)
```

1: $\int (e+f x)^m \left(a+b F^u\right)^p \left(c+d F^v\right)^q \, dx$ when $(p \mid q) \in \mathbb{Z} \land \frac{u}{v} \in \mathbb{R}$

Derivation: Algebraic expansion

Rule: If $(p \mid q) \in \mathbb{Z} \land \frac{u}{v} \in \mathbb{R}$, then

$$\int (e+f x)^m \left(a+b F^u\right)^p \left(c+d F^v\right)^q \, dx \longrightarrow \int (e+f x)^m \, \text{ExpandIntegrand}\left[\left(a+b F^u\right)^p \left(c+d F^v\right)^q, x\right] dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(a_.+b_.*F_^u_)^p_.*(c_.+d_.*F_^v_)^q_.,x_Symbol] :=
  With[{w=ExpandIntegrand[(e+f*x)^m,(a+b*F^u)^p*(c+d*F^v)^q,x]},
  Int[w,x] /;
 SumQ[w]] /;
FreeQ[{F,a,b,c,d,e,f,m},x] && IntegersQ[p,q] && LinearQ[{u,v},x] && RationalQ[Simplify[u/v]]
```

5. $\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$

1: $\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $\frac{g\,h\,\text{Log}[G]+s\,t\,\text{Log}[H]}{d\,e\,\text{Log}[F]}\in\mathbb{R}$

Derivation: Integration by substitution

Rule: If $k\in\mathbb{Z}\,\wedge\,k\,\frac{g\,h\,\text{Log}[G]+s\,t\,\text{Log}[H]}{d\,e\,\text{Log}[F]}\in\mathbb{Z}$, then

$$G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p==$$

$$\frac{k\,G^{f\,h-\frac{c\,g\,h}{d}}\,H^{r\,t-\frac{c\,s\,t}{d}}}{d\,e\,\text{Log}[F]}\,\text{Subst}\left[x^{k\,\frac{g\,h\,\text{Log}[G]+s\,t\,\text{Log}[H]}{d\,e\,\text{Log}[F]}-1}\,\left(a+b\,x^k\right)^p,\,x,\,F^{\frac{e\,(c+d\,x)}{k}}\right]\,\partial_x F^{\frac{e\,(c+d\,x)}{k}}$$

Rule: If $\frac{g\,h\,\text{Log}[G]+s\,t\,\text{Log}[H]}{d\,e\,\text{Log}[F]}\in\mathbb{R}$, then

$$\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx\,\rightarrow\,\frac{k\,G^{f\,h-\frac{c\,g\,h}{d}}\,H^{r\,t-\frac{c\,s\,t}{d}}}{d\,e\,\text{Log}[F]}\,\text{Subst}\left[\int x^{k\,\frac{g\,h\,\text{Log}[G]+s\,t\,\text{Log}[H]}{d\,e\,\text{Log}[F]}-1}\,\left(a+b\,x^k\right)^p\,dx,\,x,\,F^{\frac{e\,(c+d\,x)}{k}}\right]$$

Program code:

```
Int[G_^(h_.(f_.+g_.*x_))*H_^(t_.(r_.+s_.*x_))*(a_+b_.*F_^(e_.*(c_.+d_.*x_)))^p_.,x_Symbol] :=
  With[{m=FullSimplify[(g*h*Log[G]+s*t*Log[H])/(d*e*Log[F])]},
  Denominator[m]*G^(f*h-c*g*h/d)*H^(r*t-c*s*t/d)/(d*e*Log[F])*
    Subst[Int[x^(Numerator[m]-1)*(a+b*x^Denominator[m])^p,x],x,F^(e*(c+d*x)/Denominator[m])] /;
 RationalQ[m]] /;
FreeQ[{F,G,H,a,b,c,d,e,f,g,h,r,s,t,p},x]
```

2. $\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $\frac{g\,h\,\text{Log}[G]+s\,t\,\text{Log}[H]}{d\,e\,\text{Log}[F]}\notin\mathbb{R}$

1. $\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $p\in\mathbb{Z}$

1: $\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $d\,e\,p\,\text{Log}[F]+g\,h\,\text{Log}[G]==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $d\,e\,p\,\text{Log}[F]+g\,h\,\text{Log}[G]==0\,\land\,p\in\mathbb{Z}$, then $G^{h\,(f+g\,x)}==G^{\left(f-\frac{c\,g}{d}\right)\,h}\,\left(F^{e\,(c+d\,x)}\right)^{-p}$

Rule: If $d\,e\,p\,\text{Log}[F]+g\,h\,\text{Log}[G]==0\,\land\,p\in\mathbb{Z}$, then

$$\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx\,\longrightarrow\,G^{\left(f-\frac{c\,g}{d}\right)\,h}\int\left(F^{e\,(c+d\,x)}\right)^{-p}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx\,\longrightarrow\,G^{\left(f-\frac{c\,g}{d}\right)\,h}\int H^{t\,(r+s\,x)}\,\left(b+a\,F^{-e\,(c+d\,x)}\right)^p\,dx$$

Program code:

```
Int[G_^(h_.(f_.+g_.*x_))*H_^(t_.(r_.+s_.*x_))*(a_+b_.*F_^(e_.*(c_.+d_.*x_)))^p_.,x_Symbol] :=
  G^((f-c*g/d)*h)*Int[H^(t*(r+s*x))*(b+a*F^(-e*(c+d*x)))^p,x] /;
FreeQ[{F,G,H,a,b,c,d,e,f,g,h,r,s,t},x] && EqQ[d*e*p*Log[F]+g*h*Log[G],0] && IntegerQ[p]
```

2: $\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $p\in\mathbb{Z}^+$

Rule: If $p\in\mathbb{Z}^+$, then

$$\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx\;\longrightarrow\;\int \text{Expand}\left[G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\right]dx$$

Program code:

```
Int[G_^(h_.(f_.+g_.*x_))*H_^(t_.(r_.+s_.*x_))*(a_+b_.*F_^(e_.*(c_.+d_.*x_)))^p_.,x_Symbol] :=
  Int[Expand[G^(h*(f+g*x))*H^(t*(r+s*x))*(a+b*F^(e*(c+d*x)))^p,x],x] /;
FreeQ[{F,G,H,a,b,c,d,e,f,g,h,r,s,t},x] && IGtQ[p,0]
```

3: $\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $p\in\mathbb{Z}^-$

Rule: If $p\in\mathbb{Z}^-$, then

$$\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx\;\longrightarrow\;\frac{a^p\,G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}}{g\,h\,\text{Log}[G]+s\,t\,\text{Log}[H]}\,\text{Hypergeometric2F1}\left[-p,\,\frac{g\,h\,\text{Log}[G]+s\,t\,\text{Log}[H]}{d\,e\,\text{Log}[F]},\,\frac{g\,h\,\text{Log}[G]+s\,t\,\text{Log}[H]}{d\,e\,\text{Log}[F]}+1,\,-\frac{b}{a}\,F^{e\,(c+d\,x)}\right]$$

Program code:

```
Int[G_^(h_.(f_.+g_.*x_))*H_^(t_.(r_.+s_.*x_))*(a_+b_.*F_^(e_.*(c_.+d_.*x_)))^p_,x_Symbol] :=
  a^p*G^(h*(f+g*x))*H^(t*(r+s*x))/(g*h*Log[G]+s*t*Log[H])*
    Hypergeometric2F1[-p,(g*h*Log[G]+s*t*Log[H])/(d*e*Log[F]),(g*h*Log[G]+s*t*Log[H])/(d*e*Log[F])+1,Simplify[-b/a*F^(e*(c+d*x))]] /;
FreeQ[{F,G,H,a,b,c,d,e,f,g,h,r,s,t},x] && ILtQ[p,0]
```

2: $\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $p\notin\mathbb{Z}$

Rule: If $p\notin\mathbb{Z}$, then

$$\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx\,\to\,\frac{G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p}{\left(g\,h\,\text{Log}[G]+s\,t\,\text{Log}[H]\right)\,\left(\frac{a+b\,F^{e\,(c+d\,x)}}{a}\right)^p}$$
$$\text{Hypergeometric2F1}\left[-p,\,\frac{g\,h\,\text{Log}[G]+s\,t\,\text{Log}[H]}{d\,e\,\text{Log}[F]},\,\frac{g\,h\,\text{Log}[G]+s\,t\,\text{Log}[H]}{d\,e\,\text{Log}[F]}+1,\,-\frac{b}{a}\,F^{e\,(c+d\,x)}\right]$$

Program code:

```
Int[G_^(h_.(f_.+g_.*x_))*H_^(t_.(r_.+s_.*x_))*(a_+b_.*F_^(e_.*(c_.+d_.*x_)))^p_,x_Symbol] :=
  G^(h*(f+g*x))*H^(t*(r+s*x))*(a+b*F^(e*(c+d*x)))^p/((g*h*Log[G]+s*t*Log[H])*((a+b*F^(e*(c+d*x)))/a)^p)*
    Hypergeometric2F1[-p,(g*h*Log[G]+s*t*Log[H])/(d*e*Log[F]),(g*h*Log[G]+s*t*Log[H])/(d*e*Log[F])+1,Simplify[-b/a*F^(e*(c+d*x))]] /;
FreeQ[{F,G,H,a,b,c,d,e,f,g,h,r,s,t,p},x] && Not[IntegerQ[p]]
```

3: $\int G^{h\,u}\,H^{t\,w}\,\left(a+b\,F^{e\,v}\right)^p\,dx$ when $u==f+g\,x\,\land\,v==c+d\,x\,\land\,w==r+s\,x$

Derivation: Algebraic normalization

Rule: If $u==f+g\,x\,\land\,v==c+d\,x\,\land\,w==r+s\,x$, then

$$\int G^{h\,u}\,H^{t\,w}\,\left(a+b\,F^{e\,v}\right)^p\,dx\,\to\,\int G^{h\,(f+g\,x)}\,H^{t\,(r+s\,x)}\,\left(a+b\,F^{e\,(c+d\,x)}\right)^p\,dx$$

Program code:

```
Int[G_^(h_.u_)*H_^(t_.w_)*(a_+b_.*F_^(e_.*v_))^p_,x_Symbol] :=
  Int[G^(h*ExpandToSum[u,x])*H^(t*ExpandToSum[w,x])*(a+b*F^(e*ExpandToSum[v,x]))^p,x] /;
FreeQ[{F,G,H,a,b,e,h,t,p},x] && LinearQ[{u,v,w},x] && Not[LinearMatchQ[{u,v,w},x]]
```

6. $\int u\, F^{e\,(c+d\,x)}\,\left(a\,x^n + b\,F^{e\,(c+d\,x)}\right)^p\,dx$

1: $\int F^{e\,(c+d\,x)}\,\left(a\,x^n + b\,F^{e\,(c+d\,x)}\right)^p\,dx$ when $p \neq -1$

Derivation: Integration by parts

Basis: $F^{e\,(c+d\,x)}\,\left(a\,x^n + b\,F^{e\,(c+d\,x)}\right)^p == \partial_x \frac{\left(a\,x^n+b\,F^{e\,(c+d\,x)}\right)^{p+1}}{b\,d\,e\,(p+1)\,\mathrm{Log}[F]} - \frac{a\,n\,x^{n-1}\,\left(a\,x^n+b\,F^{e\,(c+d\,x)}\right)^p}{b\,d\,e\,\mathrm{Log}[F]}$

Rule: If $p \neq -1$, then

$$\int F^{e\,(c+d\,x)}\,\left(a\,x^n + b\,F^{e\,(c+d\,x)}\right)^p\,dx \longrightarrow \frac{\left(a\,x^n + b\,F^{e\,(c+d\,x)}\right)^{p+1}}{b\,d\,e\,(p+1)\,\mathrm{Log}[F]} - \frac{a\,n}{b\,d\,e\,\mathrm{Log}[F]}\int x^{n-1}\,\left(a\,x^n + b\,F^{e\,(c+d\,x)}\right)^p\,dx$$

Program code:

```
Int[F_^(e_.*(c_.+d_.*x_))*(a_.*x_^n_.+b_.*F_^(e_.*(c_.+d_.*x_)))^p_.,x_Symbol] :=
  (a*x^n+b*F^(e*(c+d*x)))^(p+1)/(b*d*e*(p+1)*Log[F]) -
  a*n/(b*d*e*Log[F])*Int[x^(n-1)*(a*x^n+b*F^(e*(c+d*x)))^p,x] /;
FreeQ[{F,a,b,c,d,e,n,p},x] && NeQ[p,-1]
```

2: $\int x^m F^{e\,(c+d\,x)} \left(a\,x^n + b\,F^{e\,(c+d\,x)}\right)^p dx$ when $p \neq -1$

Derivation: Integration by parts

Basis: $x^m F^{e\,(c+d\,x)} \left(a\,x^n + b\,F^{e\,(c+d\,x)}\right)^p == x^m\, \partial_x \frac{\left(a\,x^n+b\,F^{e\,(c+d\,x)}\right)^{p+1}}{b\,d\,e\,(p+1)\,Log[F]} - \frac{a\,n\,x^{m+n-1}\left(a\,x^n+b\,F^{e\,(c+d\,x)}\right)^p}{b\,d\,e\,Log[F]}$

Rule: If $p \neq -1$, then

$$\int x^m F^{c+d\,x} \left(a\,x^n + b\,F^{c+d\,x}\right)^p dx \longrightarrow$$

$$\frac{x^m \left(a\,x^n + b\,F^{e\,(c+d\,x)}\right)^{p+1}}{b\,d\,e\,(p+1)\,Log[F]} - \frac{a\,n}{b\,d\,e\,Log[F]} \int x^{m+n-1} \left(a\,x^n + b\,F^{e\,(c+d\,x)}\right)^p dx - \frac{m}{b\,d\,e\,(p+1)\,Log[F]} \int x^{m-1} \left(a\,x^n + b\,F^{e\,(c+d\,x)}\right)^{p+1} dx$$

Program code:

```
Int[x_^m_.*F_^(e_.*(c_.+d_.*x_))*(a_.*x_^n_.+b_.*F_^(e_.*(c_.+d_.*x_)))^p_.,x_Symbol] :=
  x^m*(a*x^n+b*F^(e*(c+d*x)))^(p+1)/(b*d*e*(p+1)*Log[F]) -
  a*n/(b*d*e*Log[F])*Int[x^(m+n-1)*(a*x^n+b*F^(e*(c+d*x)))^p,x] -
  m/(b*d*e*(p+1)*Log[F])*Int[x^(m-1)*(a*x^n+b*F^(e*(c+d*x)))^(p+1),x] /;
FreeQ[{F,a,b,c,d,e,m,n,p},x] && NeQ[p,-1]
```

7. $\int \frac{u\,(f + g\,x)^m}{a + b\,F^{d+e\,x} + c\,F^{2\,(d+e\,x)}} dx$ when $\sqrt{b^2 - 4\,a\,c} \neq 0 \land m \in \mathbb{Z}^+$

1: $\int \frac{(f + g\,x)^m}{a + b\,F^{d+e\,x} + c\,F^{2\,(d+e\,x)}} dx$ when $\sqrt{b^2 - 4\,a\,c} \neq 0 \land m \in \mathbb{Z}^+$

Derivation: Algebraic expansion

- **Basis:** If $q = \sqrt{b^2 - 4\,a\,c}$, then $\frac{1}{a+b\,z+c\,z^2} == \frac{2\,c}{q\,(b-q+2\,c\,z)} - \frac{2\,c}{q\,(b+q+2\,c\,z)}$

- **Rule:** If $\sqrt{b^2 - 4\,a\,c} \neq 0 \land m \in \mathbb{Z}^+$, let $q = \sqrt{b^2 - 4\,a\,c}$, then

$$\int \frac{(f+g\,x)^m}{a+b\,F^{d+e\,x}+c\,F^{2\,(d+e\,x)}}\,dx \longrightarrow \frac{2\,c}{q}\int \frac{(f+g\,x)^m}{b-q+2\,c\,F^{d+e\,x}}\,dx-\frac{2\,c}{q}\int \frac{(f+g\,x)^m}{b+q+2\,c\,F^{d+e\,x}}\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^m_./(a_.+b_.*F_^u_+c_.*F_^v_),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[(f+g*x)^m/(b-q+2*c*F^u),x] - 2*c/q*Int[(f+g*x)^m/(b+q+2*c*F^u),x]] /;
FreeQ[{F,a,b,c,f,g},x] && EqQ[v,2*u] && LinearQ[u,x] && NeQ[b^2-4*a*c,0] && IGtQ[m,0]
```

2: $\int \frac{(f+g\,x)^m\,F^{d+e\,x}}{a+b\,F^{d+e\,x}+c\,F^{2\,(d+e\,x)}}\,dx$ when $\sqrt{b^2-4\,a\,c}\neq 0\,\land\,m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

- Basis: If $q=\sqrt{b^2-4\,a\,c}$, then $\frac{1}{a+b\,z+c\,z^2}==\frac{2\,c}{q\,(b-q+2\,c\,z)}-\frac{2\,c}{q\,(b+q+2\,c\,z)}$
- Rule: If $\sqrt{b^2-4\,a\,c}\neq 0\,\land\,m\in\mathbb{Z}^+$, let $q=\sqrt{b^2-4\,a\,c}$, then

$$\int \frac{(f+g\,x)^m\,F^{d+e\,x}}{a+b\,F^{d+e\,x}+c\,F^{2\,(d+e\,x)}}\,dx \longrightarrow \frac{2\,c}{q}\int \frac{(f+g\,x)^m\,F^{d+e\,x}}{b-q+2\,c\,F^{d+e\,x}}\,dx-\frac{2\,c}{q}\int \frac{(f+g\,x)^m\,F^{d+e\,x}}{b+q+2\,c\,F^{d+e\,x}}\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*F_^u_/(a_.+b_.*F_^u_+c_.*F_^v_),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[(f+g*x)^m*F^u/(b-q+2*c*F^u),x] - 2*c/q*Int[(f+g*x)^m*F^u/(b+q+2*c*F^u),x]] /;
FreeQ[{F,a,b,c,f,g},x] && EqQ[v,2*u] && LinearQ[u,x] && NeQ[b^2-4*a*c,0] && IGtQ[m,0]
```

3: $\int \frac{(f+g\,x)^m\,\left(h+i\,F^{d+e\,x}\right)}{a+b\,F^{d+e\,x}+c\,F^{2\,(d+e\,x)}}\,dx$ when $\sqrt{b^2-4\,a\,c}\neq 0\,\wedge\,m\in\mathbb{Z}^+$

Derivation: Algebraic expansion

- Basis: If $q=\sqrt{b^2-4\,a\,c}$, then $\frac{h+i\,z}{a+b\,z+c\,z^2}==\left(\frac{2\,c\,h-b\,i}{q}+i\right)\frac{1}{b-q+2\,c\,z}-\left(\frac{2\,c\,h-b\,i}{q}-i\right)\frac{1}{b+q+2\,c\,z}$
- Rule: If $\sqrt{b^2-4\,a\,c}\neq 0\,\wedge\,m\in\mathbb{Z}^+$, let $q=\sqrt{b^2-4\,a\,c}$, then

$$\int \frac{(f+g\,x)^m\,\left(h+i\,F^{d+e\,x}\right)}{a+b\,F^{d+e\,x}+c\,F^{2\,(d+e\,x)}}\,dx \to \left(\frac{2\,c\,h-b\,i}{q}+i\right)\int \frac{(f+g\,x)^m}{b-q+2\,c\,F^{d+e\,x}}\,dx-\left(\frac{2\,c\,h-b\,i}{q}-i\right)\int \frac{(f+g\,x)^m}{b+q+2\,c\,F^{d+e\,x}}\,dx$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*(h_+i_.*F_^u_)/(a_.+b_.*F_^u_+c_.*F_^v_),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (Simplify[(2*c*h-b*i)/q]+i)*Int[(f+g*x)^m/(b-q+2*c*F^u),x] -
   (Simplify[(2*c*h-b*i)/q]-i)*Int[(f+g*x)^m/(b+q+2*c*F^u),x]] /;
FreeQ[{F,a,b,c,f,g,h,i},x] && EqQ[v,2*u] && LinearQ[u,x] && NeQ[b^2-4*a*c,0] && IGtQ[m,0]
```

8. $\int \frac{u}{a+b\,F^{d+e\,x}+c\,F^{-(d+e\,x)}}\,dx$

1: $\int \frac{x^m}{a\,F^{c+d\,x}+b\,F^{-(c+d*x)}}\,dx$ when $m>0$

Derivation: Integration by parts

Rule: If $m>0$, then

$$\int \frac{x^m}{a\,F^{c+d\,x}+b\,F^{-(c+d*x)}}\,dx \rightarrow x^m \int \frac{1}{a\,F^{c+d\,x}+b\,F^{-(c+d*x)}}\,dx - m\int x^{m-1}\int \frac{1}{a\,F^{c+d\,x}+b\,F^{-(c+d*x)}}\,dx\,dx$$

Program code:

```
Int[x_^m_./(a_.*F_^(c_.+d_.*x_)+b_.*F_^v_),x_Symbol] :=
  With[{u=IntHide[1/(a*F^(c+d*x)+b*F^v),x]},
  x^m*u - m*Int[x^(m-1)*u,x]] /;
FreeQ[{F,a,b,c,d},x] && EqQ[v,-(c+d*x)] && GtQ[m,0]
```

2: $\displaystyle\int \frac{u}{a + b\, F^{d+e\,x} + c\, F^{-(d+e\,x)}}\, dx$

Derivation: Algebraic simplification

Basis: $\frac{1}{a+b\,z+\frac{c}{z}} == \frac{z}{c+a\,z+b\,z^2}$

Rule:

$$\int \frac{u}{a + b\, F^{d+e\,x} + c\, F^{-(d+e\,x)}}\, dx \longrightarrow \int \frac{u\, F^{d+e\,x}}{c + a\, F^{d+e\,x} + b\, F^{2\,(d+e\,x)}}\, dx$$

Program code:

```
Int[u_/(a_+b_.*F_^v_+c_.*F_^w_),x_Symbol] :=
  Int[u*F^v/(c+a*F^v+b*F^(2*v)),x] /;
FreeQ[{F,a,b,c},x] && EqQ[w,-v] && LinearQ[v,x] &&
  If[RationalQ[Coefficient[v,x,1]], GtQ[Coefficient[v,x,1],0], LtQ[LeafCount[v],LeafCount[w]]]
```

9. $\int \frac{u\,F^{g\,(d+e\,x)^n}}{a+b\,x+c\,x^2}\,dx$

1: $\int \frac{F^{g\,(d+e\,x)^n}}{a+b\,x+c\,x^2}\,dx$

Derivation: Algebraic expansion

Rule:

$$\int \frac{F^{g\,(d+e\,x)^n}}{a+b\,x+c\,x^2}\,dx \longrightarrow \int F^{g\,(d+e\,x)^n}\,\text{ExpandIntegrand}\left[\frac{1}{a+b\,x+c\,x^2},\,x\right]dx$$

Program code:

```
Int[F_^(g_.*(d_.+e_.*x_)^n_.)/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[F^(g*(d+e*x)^n),1/(a+b*x+c*x^2),x],x] /;
FreeQ[{F,a,b,c,d,e,g,n},x]
```

```
Int[F_^(g_.*(d_.+e_.*x_)^n_.)/(a_+c_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[F^(g*(d+e*x)^n),1/(a+c*x^2),x],x] /;
FreeQ[{F,a,c,d,e,g,n},x]
```

2: $\int \frac{P_x^m F^{g (d+e x)^n}}{a+b x+c x^2} \, dx$

Derivation: Algebraic expansion

Rule:

$$\int \frac{P_x^m F^{g (d+e x)^n}}{a+b x+c x^2} \, dx \longrightarrow \int F^{g (d+e x)^n} \, \text{ExpandIntegrand}\left[\frac{P_x^m}{a+b x+c x^2}, x\right] dx$$

Program code:

```
Int[u_^m_.*F_^(g_.*(d_.+e_.*x_)^n_.)/(a_.+b_.*x_+c_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[F^(g*(d+e*x)^n),u^m/(a+b*x+c*x^2),x],x] /;
FreeQ[{F,a,b,c,d,e,g,n},x] && PolynomialQ[u,x] && IntegerQ[m]
```

```
Int[u_^m_.*F_^(g_.*(d_.+e_.*x_)^n_.)/(a_+c_.*x_^2),x_Symbol] :=
  Int[ExpandIntegrand[F^(g*(d+e*x)^n),u^m/(a+c*x^2),x],x] /;
FreeQ[{F,a,c,d,e,g,n},x] && PolynomialQ[u,x] && IntegerQ[m]
```

10: $\int F^{\frac{a+b x^4}{x^2}} \, dx$

Derivation: Integration by substitution

Rule:

$$\int F^{\frac{a+b x^4}{x^2}} \, dx \rightarrow$$

$$\frac{\sqrt{\pi}\, \text{Exp}\left[2 \sqrt{-a \,\text{Log}[F]} \sqrt{-b \,\text{Log}[F]}\right] \text{Erf}\left[\frac{\sqrt{-a \,\text{Log}[F]} + \sqrt{-b \,\text{Log}[F]}\, x^2}{x}\right]}{4 \sqrt{-b \,\text{Log}[F]}} - \frac{\sqrt{\pi}\, \text{Exp}\left[-2 \sqrt{-a \,\text{Log}[F]} \sqrt{-b \,\text{Log}[F]}\right] \text{Erf}\left[\frac{\sqrt{-a \,\text{Log}[F]} - \sqrt{-b \,\text{Log}[F]}\, x^2}{x}\right]}{4 \sqrt{-b \,\text{Log}[F]}}$$

Program code:

```
Int[F_^((a_.+b_.*x_^4)/x_^2),x_Symbol] :=
  Sqrt[Pi]*Exp[2*Sqrt[-a*Log[F]]*Sqrt[-b*Log[F]]]*Erf[(Sqrt[-a*Log[F]]+Sqrt[-b*Log[F]]*x^2)/x]/
    (4*Sqrt[-b*Log[F]]) -
  Sqrt[Pi]*Exp[-2*Sqrt[-a*Log[F]]*Sqrt[-b*Log[F]]]*Erf[(Sqrt[-a*Log[F]]-Sqrt[-b*Log[F]]*x^2)/x]/
    (4*Sqrt[-b*Log[F]]) /;
FreeQ[{F,a,b},x]
```

11: $\int x^m \left(e^x + x^m\right)^n dx$ when $m > 0 \land n < 0 \land n \neq -1$

Derivation: Algebraic expansion

Basis: $x^m \left(e^x + x^m\right)^n == -\left(e^x + m\,x^{m-1}\right) \left(e^x + x^m\right)^n + \left(e^x + x^m\right)^{n+1} + m\,x^{m-1} \left(e^x + x^m\right)^n$

Rule: If $m > 0 \land n < 0 \land n \neq -1$, then

$$\int x^m \left(e^x + x^m\right)^n dx \longrightarrow -\frac{\left(e^x + x^m\right)^{n+1}}{n+1} + \int \left(e^x + x^m\right)^{n+1} dx + m \int x^{m-1} \left(e^x + x^m\right)^n dx$$

Program code:

```
Int[x_^m_.*(E^x_+x_^m_.)^n_,x_Symbol] :=
  -(E^x+x^m)^(n+1)/(n+1) +
  Int[(E^x+x^m)^(n+1),x] +
  m*Int[x^(m-1)*(E^x+x^m)^n,x] /;
RationalQ[m,n] && GtQ[m,0] && LtQ[n,0] && NeQ[n,-1]
```

12: $\int u\, F^{a\,(v+b\,\text{Log}[z])}\,dx$

Derivation: Algebraic simplification

Basis: $F^{a\,(v+b\,\text{Log}[z])} == F^{a\,v}\,z^{a\,b\,\text{Log}[F]}$

Rule:

$$\int u\, F^{a\,(v+b\,\text{Log}[z])}\,dx \longrightarrow \int u\, F^{a\,v}\,z^{a\,b\,\text{Log}[F]}\,dx$$

Program code:

```
Int[u_.*F_^(a_.*(v_.+b_.*Log[z_])),x_Symbol] :=
  Int[u*F^(a*v)*z^(a*b*Log[F]),x] /;
FreeQ[{F,a,b},x]
```

13. $\int u\, F^{d\,(a+b\,\text{Log}[c\,x^n]^2)}\,dx$

1: $\int F^{d\,(a+b\,\text{Log}[c\,x^n]^2)}\,dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{x}{(c\,x^n)^{\frac{1}{n}}} == 0$

Basis: $\frac{G[\text{Log}[c\,x^n]]}{x} == \frac{1}{n}\,\text{Subst}[G[x],\,x,\,\text{Log}[c\,x^n]]\,\partial_x \text{Log}[c\,x^n]$

Rule:

$$\int F^{d\,(a+b\,\text{Log}[c\,x^n]^2)}\,dx \longrightarrow \frac{x}{(c\,x^n)^{\frac{1}{n}}}\int \frac{e^{\text{Log}[c\,x^n]/n}\,F^{d\,(a+b\,\text{Log}[c\,x^n]^2)}}{x}\,dx$$

$$\rightarrow \frac{x}{n\,(c\,x^n)^{\frac{1}{n}}}\,\text{Subst}\Big[\int e^{a\,d\,\text{Log}[F]+x/n+b\,d\,\text{Log}[F]\,x^2}\,dx,\ x,\ \text{Log}[c\,x^n]\Big]$$

Program code:

```
Int[F_^(d_.*(a_.+b_.*Log[c_.*x_^n_.]^2)),x_Symbol] :=
  x/(n*(c*x^n)^(1/n))*Subst[Int[E^(a*d*Log[F]+x/n+b*d*Log[F]*x^2),x],x,Log[c*x^n]] /;
FreeQ[{F,a,b,c,d,n},x]
```

2: $\int (e\,x)^m\,F^{d\,(a+b\,\text{Log}[c\,x^n]^2)}\,dx$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{(e\,x)^{m+1}}{(c\,x^n)^{\frac{m+1}{n}}} == 0$

Basis: $\frac{G[\text{Log}[c\,x^n]]}{x} == \frac{1}{n}\,\text{Subst}[G[x],\ x,\ \text{Log}[c\,x^n]]\ \partial_x \text{Log}[c\,x^n]$

Rule:

$$\int (e\,x)^m\,F^{d\,(a+b\,\text{Log}[c\,x^n]^2)}\,dx \rightarrow \frac{(e\,x)^{m+1}}{e\,(c\,x^n)^{\frac{m+1}{n}}}\int \frac{e^{(m+1)\,\text{Log}[c\,x^n]/n}\,F^{d\,(a+b\,\text{Log}[c\,x^n]^2)}}{x}\,dx$$

$$\rightarrow \frac{(e\,x)^{m+1}}{e\,n\,(c\,x^n)^{\frac{m+1}{n}}}\,\text{Subst}\Big[\int e^{a\,d\,\text{Log}[F]+(m+1)\,x/n+b\,d\,\text{Log}[F]\,x^2}\,dx,\ x,\ \text{Log}[c\,x^n]\Big]$$

Program code:

```
Int[(e_.*x_)^m_.*F_^(d_.*(a_.+b_.*Log[c_.*x_^n_.]^2)),x_Symbol] :=
  (e*x)^(m+1)/(e*n*(c*x^n)^((m+1)/n))*Subst[Int[E^(a*d*Log[F]+(m+1)*x/n+b*d*Log[F]*x^2),x],x,Log[c*x^n]] /;
FreeQ[{F,a,b,c,d,e,m,n},x]
```

14. $\int u\, F^{d\,(a+b\,\text{Log}[c\,x^n])^2}\, dx$

1: $\int F^{d\,(a+b\,\text{Log}[c\,x^n])^2}\, dx$

Derivation: Algebraic expansion

Rule:

$$\int F^{d\,(a+b\,\text{Log}[c\,x^n])^2}\, dx \;\longrightarrow\; \int F^{a^2\,d+2\,a\,b\,d\,\text{Log}[c\,x^n]+b^2\,d\,\text{Log}[c\,x^n]^2}\, dx$$

Program code:

```
Int[F_^(d_.*(a_.+b_.*Log[c_.*x_^n_.])^2),x_Symbol] :=
  Int[F^(a^2*d+2*a*b*d*Log[c*x^n]+b^2*d*Log[c*x^n]^2),x] /;
FreeQ[{F,a,b,c,d,n},x]
```

2: $\int (e\,x)^m\, F^{d\,(a+b\,\text{Log}[c\,x^n])^2}\, dx$

Derivation: Algebraic expansion

Rule:

$$\int (e\,x)^m\, F^{d\,(a+b\,\text{Log}[c\,x^n])^2}\, dx \;\longrightarrow\; \int (e\,x)^m\, F^{a^2\,d+2\,a\,b\,d\,\text{Log}[c\,x^n]+b^2\,d\,\text{Log}[c\,x^n]^2}\, dx$$

Program code:

```
Int[(e_.*x_)^m_.*F_^(d_.*(a_.+b_.*Log[c_.*x_^n_.])^2),x_Symbol] :=
  Int[(e*x)^m*F^(a^2*d+2*a*b*d*Log[c*x^n]+b^2*d*Log[c*x^n]^2),x] /;
FreeQ[{F,a,b,c,d,e,m,n},x]
```

15. $\int \text{Log}\left[a+b\left(F^{e\,(c+d\,x)}\right)^n\right] dx$

1: $\int \text{Log}\left[a+b\left(F^{e\,(c+d\,x)}\right)^n\right] dx$ when $a>0$

Derivation: Integration by substitution

Basis: $f\left[\left(F^{e\,(c+d\,x)}\right)^n\right] == \frac{1}{d\,e\,n\,\text{Log}[F]}\,\text{Subst}\left[\frac{f[x]}{x},\,x,\,\left(F^{e\,(c+d\,x)}\right)^n\right]\partial_x\left(F^{e\,(c+d\,x)}\right)^n$

Rule:

$$\int \text{Log}\left[a+b\left(F^{e\,(c+d\,x)}\right)^n\right] dx \longrightarrow \frac{1}{d\,e\,n\,\text{Log}[F]}\,\text{Subst}\left[\int\frac{\text{Log}[a+b\,x]}{x}\,dx,\,x,\,\left(F^{e\,(c+d\,x)}\right)^n\right]$$

Program code:

```
Int[Log[a_+b_.*(F_^(e_.*(c_.+d_.*x_)))^n_.],x_Symbol] :=
  1/(d*e*n*Log[F])*Subst[Int[Log[a+b*x]/x,x],x,(F^(e*(c+d*x)))^n] /;
FreeQ[{F,a,b,c,d,e,n},x] && GtQ[a,0]
```

2: $\int \text{Log}\left[a+b\left(F^{e\,(c+d\,x)}\right)^n\right] dx$ when $a \not> 0$

Derivation: Integration by parts

Rule: If $a \not> 0$, then

$$\int \text{Log}\left[a+b\left(F^{e\,(c+d\,x)}\right)^n\right] dx \longrightarrow x\,\text{Log}\left[a+b\left(F^{e\,(c+d\,x)}\right)^n\right] - b\,d\,e\,n\,\text{Log}[F] \int \frac{x\left(F^{e\,(c+d\,x)}\right)^n}{a+b\left(F^{e\,(c+d\,x)}\right)^n}\,dx$$

Program code:

```
Int[Log[a_+b_.*(F_^(e_.*(c_.+d_.*x_)))^n_.],x_Symbol] :=
  x*Log[a+b*(F^(e*(c+d*x)))^n] - b*d*e*n*Log[F]*Int[x*(F^(e*(c+d*x)))^n/(a+b*(F^(e*(c+d*x)))^n),x] /;
FreeQ[{F,a,b,c,d,e,n},x] && Not[GtQ[a,0]]
```

16. $\int u\,(a\,F^{v})^{n}\,dx$

x: $\int u\,(a\,F^{v})^{n}\,dx$ when $n \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $n \in \mathbb{Z}$, then $(a\,F^{v})^{n} == a^{n}\,F^{n\,v}$

Note: This rule not necessary since *Mathematica* automatically does this simplification.

Rule: If $n \in \mathbb{Z}$, then

$$\int u\,(a\,F^{v})^{n}\,dx \longrightarrow a^{n}\int u\,F^{n\,v}\,dx$$

Program code:

```
(* Int[u_.*(a_.*F_^v_)^n_,x_Symbol] :=
  a^n*Int[u*F^(n*v),x] /;
FreeQ[{F,a},x] && IntegerQ[n] *)
```

2: $\int u\,(a\,F^v)^n\,dx$ when $n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a\,F^{v[x]})^n}{F^{n\,v[x]}} == 0$

Rule: If $n \notin \mathbb{Z}$, then

$$\int u\,(a\,F^v)^n\,dx \longrightarrow \frac{(a\,F^v)^n}{F^{n\,v}} \int u\,F^{n\,v}\,dx$$

Program code:

```
Int[u_.*(a_.*F_^v_)^n_,x_Symbol] :=
  (a*F^v)^n/F^(n*v)*Int[u*F^(n*v),x] /;
FreeQ[{F,a,n},x] && Not[IntegerQ[n]]
```

17: $\int f\left[F^{a+b\,x}\right] dx$

Derivation: Integration by substitution

Basis: $f\left[F^{a+b\,x}\right] = \frac{1}{b\,\text{Log}[F]}\,\text{Subst}\left[\frac{f[x]}{x},\,x,\,F^{a+b\,x}\right]\partial_x F^{a+b\,x}$

Basis: $\frac{1}{b\,\text{Log}[F]} = \frac{F^{a+b\,x}}{\partial_x F^{a+b\,x}}$

Rule:

$$\int f\left[F^{a+b\,x}\right] dx \longrightarrow \frac{F^{a+b\,x}}{\partial_x F^{a+b\,x}}\,\text{Subst}\left[\int \frac{f[x]}{x}\,dx,\,x,\,F^{a+b\,x}\right]$$

Program code:

```
Int[u_,x_Symbol] :=
  With[{v=FunctionOfExponential[u,x]},
  v/D[v,x]*Subst[Int[FunctionOfExponentialFunction[u,x]/x,x],x,v]] /;
FunctionOfExponentialQ[u,x]
```

18. $\int u\,(a\,F^v + b\,G^w)^n\,dx$

1. $\int u\,(a\,F^v + b\,G^w)^n\,dx$ when $n \in \mathbb{Z}^-$

1: $\int u\,(a\,F^v + b\,F^w)^n\,dx$ when $n \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Rule: If $n \in \mathbb{Z}^-$, then

$$\int u\,(a\,F^v + b\,F^w)^n\,dx \longrightarrow \int u\,F^{n\,v}\,(a + b\,F^{w-v})^n\,dx$$

Program code:

```
Int[u_.*(a_.*F_^v_+b_.*F_^w_)^n_,x_Symbol] :=
  Int[u*F^(n*v)*(a+b*F^ExpandToSum[w-v,x])^n,x] /;
FreeQ[{F,a,b,n},x] && ILtQ[n,0] && LinearQ[{v,w},x]
```

2: $\int u\,(a\,F^v + b\,G^w)^n\,dx$ when $n \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Rule: If $n \in \mathbb{Z}^-$, then

$$\int u\,(a\,F^v + b\,G^w)^n\,dx \longrightarrow \int u\,F^{n\,v}\,\left(a + b\,E^{\mathrm{Log}[G]\,w-\mathrm{Log}[F]\,v}\right)^n\,dx$$

Program code:

```
Int[u_.*(a_.*F_^v_+b_.*G_^w_)^n_,x_Symbol] :=
  Int[u*F^(n*v)*(a+b*E^ExpandToSum[Log[G]*w-Log[F]*v,x])^n,x] /;
FreeQ[{F,G,a,b,n},x] && ILtQ[n,0] && LinearQ[{v,w},x]
```

2. $\int u \left(a F^v + b G^w\right)^n dx$ when $n \notin \mathbb{Z}$

1: $\int u \left(a F^v + b F^w\right)^n dx$ when $n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\left(a F^{f[x]}+b F^{g[x]}\right)^n}{F^{n f[x]} \left(a+b F^{g[x]-f[x]}\right)^n} == 0$

Rule: If $n \notin \mathbb{Z}$, then

$$\int u \left(a F^v + b F^w\right)^n dx \longrightarrow \frac{\left(a F^v + b F^w\right)^n}{F^{n v} \left(a + b F^{w-v}\right)^n} \int u F^{n v} \left(a + b F^{w-v}\right)^n dx$$

Program code:

```
Int[u_.*(a_.*F_^v_+b_.*F_^w_)^n_,x_Symbol] :=
  (a*F^v+b*F^w)^n/(F^(n*v)*(a+b*F^ExpandToSum[w-v,x])^n)*Int[u*F^(n*v)*(a+b*F^ExpandToSum[w-v,x])^n,x] /;
FreeQ[{F,a,b,n},x] && Not[IntegerQ[n]] && LinearQ[{v,w},x]
```

2: $\int u \left(a F^v + b G^w\right)^n dx$ when $n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\left(a F^{f[x]}+b G^{g[x]}\right)^n}{F^{n f[x]} \left(a+b E^{Log[G] g[x]-Log[F] f[x]}\right)^n} == 0$

Rule: If $n \notin \mathbb{Z}$, then

$$\int u \left(a F^v + b G^w\right)^n dx \to \frac{\left(a F^v + b G^w\right)^n}{F^{n v} \left(a + b E^{Log[G] w-Log[F] v}\right)^n} \int u F^{n v} \left(a + b E^{Log[G] w-Log[F] v}\right)^n dx$$

Program code:

```
Int[u_.*(a_.*F_^v_+b_.*G_^w_)^n_,x_Symbol] :=
  (a*F^v+b*G^w)^n/(F^(n*v)*(a+b*E^ExpandToSum[Log[G]*w-Log[F]*v,x])^n)*Int[u*F^(n*v)*(a+b*E^ExpandToSum[Log[G]*w-Log[F]*v,x])^n,x] /;
FreeQ[{F,G,a,b,n},x] && Not[IntegerQ[n]] && LinearQ[{v,w},x]
```

19: $\int u\, F^v\, G^w\, dx$

Derivation: Algebraic simplification

Basis: $F^v\, G^w == E^{v\, \text{Log}[F]+w\, \text{Log}[G]}$

Rule:

$$\int u\, F^v\, G^w\, dx \longrightarrow \int u\, E^{v\, \text{Log}[F]+w\, \text{Log}[G]}\, dx$$

Program code:

```
Int[u_.*F_^v_*G_^w_,x_Symbol] :=
  With[{z=v*Log[F]+w*Log[G]},
  Int[u*NormalizeIntegrand[E^z,x],x] /;
 BinomialQ[z,x] || PolynomialQ[z,x] && LeQ[Exponent[z,x],2]] /;
FreeQ[{F,G},x]
```

20: $\int F^u (v + w)\, y \, dx$ when $\partial_x \frac{v\, y}{\mathrm{Log}[F]\, \partial_x u} == w\, y$

Basis: $\partial_x \left(F^{f[x]}\, g[x]\right) == F^{f[x]}\, (\mathrm{Log}[F]\, g[x]\, f'[x] + g'[x])$

Rule: Let $z = \frac{v\, y}{\mathrm{Log}[F]\, \partial_x u}$, if $\partial_x z == w\, y$, then

$$\int F^u (v + w)\, y \, dx \longrightarrow F^{f[x]}\, z$$

Program code:

```
Int[F_^u_*(v_+w_)*y_.,x_Symbol] :=
  With[{z=v*y/(Log[F]*D[u,x])},
  F^u*z /;
 EqQ[D[z,x],w*y]] /;
FreeQ[F,x]
```

21: $\int F^u v^n w \, dx$ when $\text{Log}[F] \, v \, \partial_x u + (n+1) \, \partial_x v$ divides w

Basis: $\partial_x \left(F^{f[x]} g[x]^{n+1}\right) == F^{f[x]} g[x]^n \left(\text{Log}[F] \, g[x] \, f'[x] + (n+1) \, g'[x]\right)$

Rule: Let $z = \text{Log}[F] \, v \, \partial_x u + (n+1) \, \partial_x v$, if z divides w, then

$$\int F^u v^n w \, dx \to \frac{w}{z} F^u v^{n+1}$$

Program code:

```
Int[F_^u_*v_^n_.*w_,x_Symbol] :=
  With[{z=Log[F]*v*D[u,x]+(n+1)*D[v,x]},
  Coefficient[w,x,Exponent[w,x]]/Coefficient[z,x,Exponent[z,x]]*F^u*v^(n+1) /;
 EqQ[Exponent[w,x],Exponent[z,x]] && EqQ[w*Coefficient[z,x,Exponent[z,x]],z*Coefficient[w,x,Exponent[w,x]]]] /;
FreeQ[{F,n},x] && PolynomialQ[u,x] && PolynomialQ[v,x] && PolynomialQ[w,x]
```

22. $\int u \frac{\left(a + b F^{c \frac{\sqrt{d+e x}}{\sqrt{f+g x}}}\right)^n}{A + B x + C x^2} dx$ when $C d f - A e g == 0 \land B e g - C (e f + d g) == 0$

1: $\int \frac{\left(a + b F^{c \frac{\sqrt{d+e x}}{\sqrt{f+g x}}}\right)^n}{A + B x + C x^2} dx$ when $C d f - A e g == 0 \land B e g - C (e f + d g) == 0 \land n \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x] == 2 (e f - d g) \, \text{Subst}\left[\frac{x}{(e - g x^2)^2} F\left[-\frac{d - f x^2}{e - g x^2}\right], x, \frac{\sqrt{d+e x}}{\sqrt{f+g x}}\right] \partial_x \frac{\sqrt{d+e x}}{\sqrt{f+g x}}$

Basis: If $C d f - A e g == 0 \land B e g - C (e f + d g) == 0$, then

$$\frac{1}{A+B\,x+C\,x^2} = \frac{2\,e\,g}{C\,(e\,f-d\,g)}\,\text{Subst}\left[\frac{1}{x},\,x,\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]\partial_x\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}$$

Rule: If $C\,d\,f-A\,e\,g=0\,\land\,B\,e\,g-C\,(e\,f+d\,g)=0\,\land\,n\in\mathbb{Z}^+$, then

$$\int\frac{\left(a+b\,F^{c\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}}\right)^n}{A+B\,x+C\,x^2}\,dx\,\longrightarrow\,\frac{2\,e\,g}{C\,(e\,f-d\,g)}\,\text{Subst}\left[\int\frac{\left(a+b\,F^{c\,x}\right)^n}{x}\,dx,\,x,\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]$$

Program code:

```
Int[(a_.+b_.*F_^(c_.*Sqrt[d_.+e_.*x_]/Sqrt[f_.+g_.*x_]))^n_./(A_.+B_.*x_+C_.*x_^2),x_Symbol] :=
  2*e*g/(C*(e*f-d*g))*Subst[Int[(a+b*F^(c*x))^n/x,x],x,Sqrt[d+e*x]/Sqrt[f+g*x]] /;
FreeQ[{a,b,c,d,e,f,g,A,B,C,F},x] && EqQ[C*d*f-A*e*g,0] && EqQ[B*e*g-C*(e*f+d*g),0] && IGtQ[n,0]
```

```
Int[(a_.+b_.*F_^(c_.*Sqrt[d_.+e_.*x_]/Sqrt[f_.+g_.*x_]))^n_./(A_+C_.*x_^2),x_Symbol] :=
  2*e*g/(C*(e*f-d*g))*Subst[Int[(a+b*F^(c*x))^n/x,x],x,Sqrt[d+e*x]/Sqrt[f+g*x]] /;
FreeQ[{a,b,c,d,e,f,g,A,C,F},x] && EqQ[C*d*f-A*e*g,0] && EqQ[e*f+d*g,0] && IGtQ[n,0]
```

2: $\int \frac{\left(a+b\,F^{\frac{c\sqrt{d+e\,x}}{\sqrt{f+g\,x}}}\right)^n}{A+B\,x+C\,x^2}\,dx$ when $C\,d\,f-A\,e\,g==0\,\land\,B\,e\,g-C\,(e\,f+d\,g)==0\,\land\,n\notin\mathbb{Z}^+$

Rule: If $C\,d\,f-A\,e\,g==0\,\land\,B\,e\,g-C\,(e\,f+d\,g)==0\,\land\,n\notin\mathbb{Z}^+$, then

$$\int \frac{\left(a+b\,F^{\frac{c\sqrt{d+e\,x}}{\sqrt{f+g\,x}}}\right)^n}{A+B\,x+C\,x^2}\,dx \longrightarrow \int \frac{\left(a+b\,F^{\frac{c\sqrt{d+e\,x}}{\sqrt{f+g\,x}}}\right)^n}{A+B\,x+C\,x^2}\,dx$$

Program code:

```
Int[(a_.+b_.*F_^(c_.*Sqrt[d_.+e_.*x_]/Sqrt[f_.+g_.*x_]))^n_/(A_.+B_.*x_+C_.*x_^2),x_Symbol] :=
  Unintegrable[(a+b*F^(c*Sqrt[d+e*x]/Sqrt[f+g*x]))^n/(A+B*x+C*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g,A,B,C,F,n},x] && EqQ[C*d*f-A*e*g,0] && EqQ[B*e*g-C*(e*f+d*g),0] && Not[IGtQ[n,0]]
```

```
Int[(a_.+b_.*F_^(c_.*Sqrt[d_.+e_.*x_]/Sqrt[f_.+g_.*x_]))^n_/(A_+C_.*x_^2),x_Symbol] :=
  Unintegrable[(a+b*F^(c*Sqrt[d+e*x]/Sqrt[f+g*x]))^n/(A+C*x^2),x] /;
FreeQ[{a,b,c,d,e,f,g,A,C,F,n},x] && EqQ[C*d*f-A*e*g,0] && EqQ[e*f+d*g,0] && Not[IGtQ[n,0]]
```