

Rules for integrands of the form $P_q[x] (a + b x + c x^2)^p$ when $q > 1$

1: $\int P_q[x] (a + b x + c x^2)^p dx$ when $p + 2 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.1.8.1: If $p + 2 \in \mathbb{Z}^+$, then

$$\int P_q[x] (a + b x + c x^2)^p dx \rightarrow \int \text{ExpandIntegrand}[P_q[x] (a + b x + c x^2)^p, x] dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[ExpandIntegrand[Pq*(a+b*x+c*x^2)^p,x],x] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x] && IGtQ[p,-2]
```

2: $\int P_q[x] (a + b x + c x^2)^p dx$ when $\text{PolynomialRemainder}[P_q[x], x, x] == 0$

Derivation: Algebraic simplification

Rule 1.2.1.8.2: If $\text{PolynomialRemainder}[P_q[x], x, x] == 0$, then

$$\int P_q[x] (a + b x + c x^2)^p dx \rightarrow \int x \text{PolynomialQuotient}[P_q[x], x, x] (a + b x + c x^2)^p dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  Int[x*PolynomialQuotient[Pq,x,x]*(a+b*x+c*x^2)^p,x] /;
FreeQ[{a,b,c,p},x] && PolyQ[Pq,x] && EqQ[PolynomialRemainder[Pq,x,x],0] && Not[MatchQ[Pq,x^m_.*u_./; IntegerQ[m]]]
```

3: $\int P_q[x] (a+bx+cx^2)^p dx$ when $b^2 - 4ac = 0$

Derivation: Piecewise constant extraction

Basis: If $b^2 - 4ac = 0$, then $\partial_x \frac{(a+bx+cx^2)^p}{(b+2cx)^{2p}} = 0$

Rule 1.2.1.8.3: If $b^2 - 4ac = 0$, then

$$\int P_q[x] (a+bx+cx^2)^p dx \rightarrow \frac{(a+bx+cx^2)^{\text{FracPart}[p]}}{(4c)^{\text{IntPart}[p]} (b+2cx)^{2\text{FracPart}[p]}} \int P_q[x] (b+2cx)^{2p} dx$$

Program code:

```
Int[Pq*(a+b_.*x+c_.*x_^2)^p_,x_Symbol] :=
  (a+b*x+c*x^2)^FracPart[p]/((4*c)^IntPart[p]*(b+2*c*x)^(2*FracPart[p]))*Int[Pq*(b+2*c*x)^(2*p),x] /;
FreeQ[{a,b,c,p},x] && PolyQ[Pq,x] && EqQ[b^2-4*a*c,0]
```

4: $\int P_q[x] (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge p < -1$

Derivation: Algebraic expansion and quadratic recurrence 2a

Rule 1.2.1.8.4: If $b^2 - 4ac \neq 0 \wedge p < -1$,

let $Q_{q-2}[x] \rightarrow \text{PolynomialQuotient}[P_q[x], a+bx+cx^2, x]$ and

$f+gx \rightarrow \text{PolynomialRemainder}[P_q[x], a+bx+cx^2, x]$, then

$$\int P_q[x] (a+bx+cx^2)^p dx \rightarrow$$

$$\int (f+gx) (a+bx+cx^2)^p dx + \int Q_{q-2}[x] (a+bx+cx^2)^{p+1} dx \rightarrow$$

$$\frac{(bf - 2ag + (2cf - bg)x)(a + bx + cx^2)^{p+1}}{(p+1)(b^2 - 4ac)} + \frac{1}{(p+1)(b^2 - 4ac)} \int (a + bx + cx^2)^{p+1} ((p+1)(b^2 - 4ac)Q_{q-2}[x] - (2p+3)(2cf - bg)) dx$$

Program code:

```
Int[Pq*(a_.+b_.*x_.+c_.*x_^2)^p_,x_Symbol] :=
  With[{Q=PolynomialQuotient[Pq,a+b*x+c*x^2,x],
    f=Coeff[PolynomialRemainder[Pq,a+b*x+c*x^2,x],x,0],
    g=Coeff[PolynomialRemainder[Pq,a+b*x+c*x^2,x],x,1]},
    (b*f-2*a*g+(2*c*f-b*g)*x)*(a+b*x+c*x^2)^(p+1)/((p+1)*(b^2-4*a*c)) +
    1/((p+1)*(b^2-4*a*c))*Int[(a+b*x+c*x^2)^(p+1)*ExpandToSum[(p+1)*(b^2-4*a*c)*Q-(2*p+3)*(2*c*f-b*g),x],x] /;
  FreeQ[{a,b,c},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && LtQ[p,-1]
```

5: $\int P_q[x] (a+bx+cx^2)^p dx$ when $b^2 - 4ac \neq 0 \wedge p \neq -1$

Reference: G&R 2.160.3

Derivation: Trinomial recurrence 3a with $A = 0$, $B = 1$ and $m = m - n$

Reference: G&R 2.104

Note: This special case of the Ostrogradskiy-Hermite integration method reduces the degree of the polynomial in the resulting integrand.

Rule 1.2.1.8.5: If $b^2 - 4ac \neq 0 \wedge p \neq -1$, let $e \rightarrow P_q[x, q]$, then

$$\begin{aligned} & \int P_q[x] (a+bx+cx^2)^p dx \rightarrow \\ & \int (P_q[x] - e x^q) (a+bx+cx^2)^p dx + e \int x^q (a+bx+cx^2)^p dx \rightarrow \\ & \frac{e x^{q-1} (a+bx+cx^2)^{p+1}}{c(q+2p+1)} + \frac{1}{c(q+2p+1)} \int (a+bx+cx^2)^p (c(q+2p+1)P_q[x] - a e (q-1)x^{q-2} - b e (q+p)x^{q-1} - c e (q+2p+1)x^q) dx \end{aligned}$$

Program code:

```
Int[Pq*(a_.+b_.*x_.+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x],e=Coeff[Pq,x,Expon[Pq,x]]},
    e*x^(q-1)*(a+b*x+c*x^2)^(p+1)/(c*(q+2*p+1)) +
    1/(c*(q+2*p+1))*Int[(a+b*x+c*x^2)^p*
      ExpandToSum[c*(q+2*p+1)*Pq-a*e*(q-1)*x^(q-2)-b*e*(q+p)*x^(q-1)-c*e*(q+2*p+1)*x^q,x],x] /;
    FreeQ[{a,b,c,p},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && Not[LeQ[p,-1]]
```