Rules for integrands of the form $u \operatorname{Log}[e (f (a+b x)^p (c+d x)^q)^r]^s$

1: $\int u \operatorname{Log}[e (f (a+b x)^p (c+d x)^q)^r]^s \, dx$ when $b c - a d == 0 \wedge p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b c - a d == 0$, then $a + b x == \frac{b}{d} (c + d x)$

Rule: If $b c - a d == 0 \wedge p \in \mathbb{Z}$, then

$$\int u \operatorname{Log}[e (f (a+b x)^p (c+d x)^q)^r]^s \, dx \longrightarrow \int u \operatorname{Log}\left[e \left(\frac{b^p f}{d^p} (c+d x)^{p+q}\right)^r\right]^s dx$$

Program code:

```
Int[u_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  Int[u*Log[e*(b^p*f/d^p*(c+d*x)^(p+q))^r]^s,x] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && EqQ[b*c-a*d,0] && IntegerQ[p]
```

2: $\int \text{Log}\left[e\left(f(a+b\,x)^p(c+d\,x)^q\right)^r\right]^s dx$ when $b\,c-a\,d\neq 0\wedge s\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: $1==\partial_x\frac{a+b\,x}{b}$

Basis: $\partial_x \text{Log}\left[e\left(f(a+b\,x)^p(c+d\,x)^q\right)^r\right]^s==\frac{r\,s\,(b\,c\,p+a\,d\,q+b\,d\,(p+q)\,x)\,\text{Log}\left[e\left(f(a+b\,x)^p(c+d\,x)^q\right)^r\right]^{s-1}}{(a+b\,x)\,(c+d\,x)}$

Rule: If $b\,c-a\,d\neq 0\wedge s\in\mathbb{Z}^+$, then

$$\int \text{Log}\left[e\left(f(a+b\,x)^p(c+d\,x)^q\right)^r\right]^s dx\longrightarrow$$

$$\frac{(a+b\,x)\,\text{Log}\left[e\left(f(a+b\,x)^p(c+d\,x)^q\right)^r\right]^s}{b}-\frac{r\,s}{b}\int\frac{(b\,c\,p+a\,d\,q+b\,d\,(p+q)\,x)\,\text{Log}\left[e\left(f(a+b\,x)^p(c+d\,x)^q\right)^r\right]^{s-1}}{c+d\,x}dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  (a+b*x)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/b -
  r*s/b*Int[(b*c*p+a*d*q+b*d*(p+q)*x)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0]
```

3. $\int (g+h\,x)^m \,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s\,dx$ when $b\,c-a\,d\neq 0$

1. $\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{g+h\,x}\,dx$ when $b\,c-a\,d\neq 0\,\land\,s\in\mathbb{Z}^+$

1: $\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{g+h\,x}\,dx$ when $b\,c-a\,d\neq 0\,\land\,s\in\mathbb{Z}^+\land\,b\,g-a\,h=0\,\land\,p+q=0$

Derivation: Integration by parts

Basis: If $b\,g-a\,h=0$, then $\frac{1}{g+h\,x}=-\frac{1}{h}\,\partial_x\,\text{Log}\left[-\frac{b\,c-a\,d}{d\,(a+b\,x)}\right]$

Basis: If $p+q=0$, then $\partial_x\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s=\frac{p\,r\,s\,(b\,c-a\,d)}{(a+b\,x)\,(c+d\,x)}\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}$

Rule: If $b\,c-a\,d\neq 0\,\land\,s\in\mathbb{Z}^+\land\,b\,g-a\,h=0\,\land\,p+q=0$, then

$$\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{g+h\,x}\,dx\,\longrightarrow$$

$$-\frac{\text{Log}\left[-\frac{b\,c-a\,d}{d\,(a+b\,x)}\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{h}+\frac{p\,r\,s\,(b\,c-a\,d)}{h}\int \frac{\text{Log}\left[-\frac{b\,c-a\,d}{d\,(a+b\,x)}\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{(a+b\,x)\,(c+d\,x)}\,dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_./(g_.+h_.*x_),x_Symbol] :=
  -Log[-(b*c-a*d)/(d*(a+b*x))]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/h +
  p*r*s(b*c-a*d)/h*Int[Log[-(b*c-a*d)/(d*(a+b*x))]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((a+b*x)*(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && EqQ[b*g-a*h,0] && EqQ[p+q,0]
```

2. $\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{g+h\,x}\,dx$ when $b\,c-a\,d\neq 0\,\land\,s\in\mathbb{Z}^+$

1: $\int \frac{\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{g+h\,x}\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Integration by parts

Basis: $\frac{1}{g+h\,x}=\partial_x\frac{\text{Log}[g+h\,x]}{h}$

Basis: $\partial_x\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]=\frac{b\,p\,r}{a+b\,x}+\frac{d\,q\,r}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0$, then

$$\int \frac{\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{g+h\,x}\,dx\to$$

$$\frac{\text{Log}[g+h\,x]\,\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}{h}-\frac{b\,p\,r}{h}\int\frac{\text{Log}[g+h\,x]}{a+b\,x}\,dx-\frac{d\,q\,r}{h}\int\frac{\text{Log}[g+h\,x]}{c+d\,x}\,dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]/(g_.+h_.*x_),x_Symbol] :=
  Log[g+h*x]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]/h -
  b*p*r/h*Int[Log[g+h*x]/(a+b*x),x] -
  d*q*r/h*Int[Log[g+h*x]/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r},x] && NeQ[b*c-a*d,0]
```

2: $\int \frac{\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{g+h\,x}\,dx$ when $b\,c-a\,d\neq 0\land s-1\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{1}{g+h\,x}==\frac{d}{h\,(c+d\,x)}-\frac{d\,g-c\,h}{h\,(c+d\,x)\,(g+h\,x)}$

Rule: If $b\,c-a\,d\neq 0\land s-1\in\mathbb{Z}^+$, then

$$\int \frac{\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{g+h\,x}\,dx \longrightarrow$$

$$\frac{d}{h}\int \frac{\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{c+d\,x}\,dx-\frac{d\,g-c\,h}{h}\int \frac{\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(c+d\,x)\,(g+h\,x)}\,dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_/(g_.+h_.*x_),x_Symbol] :=
  d/h*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(c+d*x),x] -
  (d*g-c*h)/h*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/((c+d*x)*(g+h*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,1]
```

2. $\int (g+h\,x)^m \operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,s\in\mathbb{Z}^+\wedge\,m\neq -1$

   1. $\int (g+h\,x)^m \operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,s\in\mathbb{Z}^+\wedge\,m\neq -1\,\wedge\,p+q=0$

      1: $\int \frac{\operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(g+h\,x)^2}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,s\in\mathbb{Z}^+\wedge\,p+q=0\,\wedge\,b\,g-a\,h\neq 0$

Derivation: Integration by parts

Basis: $\frac{1}{(g+h\,x)^2}=\partial_x\frac{a+b\,x}{(b\,g-a\,h)\,(g+h\,x)}$

Basis: If $p+q=0$, then $\partial_x \operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s=\frac{p\,r\,s\,(b\,c-a\,d)}{(a+b\,x)\,(c+d\,x)}\operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,s\in\mathbb{Z}^+\wedge\,p+q=0\,\wedge\,b\,g-a\,h\neq 0$, then

$$\int \frac{\operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(g+h\,x)^2}\,dx\,\longrightarrow$$

$$\frac{(a+b\,x)\operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(b\,g-a\,h)\,(g+h\,x)}-\frac{p\,r\,s\,(b\,c-a\,d)}{(b\,g-a\,h)}\int\frac{\operatorname{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{(c+d\,x)\,(g+h\,x)}\,dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_./(g_.+h_.*x_)^2,x_Symbol] :=
  (a+b*x)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/((b*g-a*h)*(g+h*x)) -
  p*r*s*(b*c-a*d)/(b*g-a*h)*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((c+d*x)*(g+h*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && EqQ[p+q,0] && NeQ[b*g-a*h,0]
```

2: $\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(g+h\,x)^3}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,s\in\mathbb{Z}^+\wedge\,p+q==0\,\wedge\,b\,g-a\,h==0\,\wedge\,d\,g-c\,h\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{(g+h\,x)^3}==\frac{d}{(d\,g-c\,h)\,(g+h\,x)^2}-\frac{h\,(c+d\,x)}{(d\,g-c\,h)\,(g+h\,x)^3}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,s\in\mathbb{Z}^+\wedge\,p+q==0\,\wedge\,b\,g-a\,h==0\,\wedge\,d\,g-c\,h\neq 0$, then

$$\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(g+h\,x)^3}\,dx\;\longrightarrow$$
$$\frac{d}{d\,g-c\,h}\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(g+h\,x)^2}\,dx-\frac{h}{d\,g-c\,h}\int \frac{(c+d\,x)\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(g+h\,x)^3}\,dx$$

Program code:

```
Int[Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_/(g_.+h_.*x_)^3,x_Symbol] :=
  d/(d*g-c*h)*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(g+h*x)^2,x] -
  h/(d*g-c*h)*Int[(c+d*x)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(g+h*x)^3,x] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && EqQ[p+q,0] && EqQ[b*g-a*h,0] && NeQ[d*g-c*h,0]
```

3: $\int (g+h\,x)^m\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,s\in\mathbb{Z}^+\wedge\,m\neq -1\,\wedge\,p+q==0$

Derivation: Integration by parts

Basis: $(g+h\,x)^m==\partial_x\frac{(g+h\,x)^{m+1}}{h\,(m+1)}$

Basis: If $p+q==0$, then $\partial_x\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s==\frac{p\,r\,s\,(b\,c-a\,d)}{(a+b\,x)\,(c+d\,x)}\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,s\in\mathbb{Z}^+\wedge\,m\neq -1\,\wedge\,p+q==0$, then

$$\int (g+h\,x)^m\,\mathrm{Log}\!\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s\,dx\;\longrightarrow$$

$$\frac{(g+h\,x)^{m+1}\,\mathrm{Log}\!\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{h\,(m+1)}-\frac{p\,r\,s\,(b\,c-a\,d)}{h\,(m+1)}\int\frac{(g+h\,x)^{m+1}\,\mathrm{Log}\!\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{(a+b\,x)\,(c+d\,x)}\,dx$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  (g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(h*(m+1)) -
  p*r*s*(b*c-a*d)/(h*(m+1))*Int[(g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((a+b*x)*(c+d*x)),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && NeQ[m,-1] && EqQ[p+q,0]
```

2: $\int (g+h x)^m \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx$ when $b c-a d \neq 0 \land s \in \mathbb{Z}^+ \land m \neq -1 \land p+q \neq 0$

Derivation: Integration by parts

Basis: $(g+h x)^m == \partial_x \frac{(g+h x)^{m+1}}{h (m+1)}$

Basis: $\partial_x \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s == \frac{b p r s}{a+b x} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1} + \frac{d q r s}{c+d x} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1}$

Rule: If $b c-a d \neq 0 \land s \in \mathbb{Z}^+ \land m \neq -1 \land p+q \neq 0$, then

$$\int (g+h x)^m \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx \longrightarrow$$

$$\frac{(g+h x)^{m+1} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{h (m+1)} - \frac{b p r s}{h (m+1)} \int \frac{(g+h x)^{m+1} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1}}{a+b x} dx - \frac{d q r s}{h (m+1)} \int \frac{(g+h x)^{m+1} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1}}{c+d x} dx$$

Program code:

```
Int[(g_.+h_.*x_)^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  (g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(h*(m+1)) -
  b*p*r*s/(h*(m+1))*Int[(g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/(a+b*x),x] -
  d*q*r*s/(h*(m+1))*Int[(g+h*x)^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,g,h,m,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && NeQ[m,-1] && NeQ[p+q,0]
```

3: $\int \frac{1}{(g+h x)^2 \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]} dx$ when $b c-a d \neq 0 \land p+q == 0 \land b g-a h == 0$

Rule: If $b c-a d \neq 0 \land p+q == 0 \land b g-a h == 0$, then

$$\int \frac{1}{(g+h\,x)^2\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]}\,dx \longrightarrow$$

$$\frac{b\,(c+d\,x)\,\left(e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right)^{\frac{1}{p\,r}}}{h\,p\,r\,(b\,c-a\,d)\,(g+h\,x)}\,\text{ExpIntegralEi}\left[-\frac{1}{p\,r}\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]\right]$$

Program code:

```
Int[1/((g_.+h_.*x_)^2*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]),x_Symbol] :=
  b*(c+d*x)*(e*(f*(a+b*x)^p*(c+d*x)^q)^r)^(1/(p*r))/(h*p*r*(b*c-a*d)*(g+h*x))*
    ExpIntegralEi[-1/(p*r)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]] /;
FreeQ[{a,b,c,d,e,f,g,h,p,q,r},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && EqQ[b*g-a*h,0]
```

4. $\int \frac{\text{Log}\left[i\,(j\,(h\,x)^t)^u\right]^m \text{Log}\left[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r\right]^s}{x}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{\text{Log}\left[i\,(j\,(h\,x)^t)^u\right]^m \text{Log}\left[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r\right]}{x}\,dx$ when $b\,c-a\,d\neq 0\wedge m\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\frac{\text{Log}\left[i\,(j\,(h\,x)^t)^u\right]^m}{x} == \partial_x \frac{\text{Log}\left[i\,(j\,(h\,x)^t)^u\right]^{m+1}}{t\,u\,(m+1)}$

Basis: $\partial_x \text{Log}\left[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r\right] == \frac{b\,p\,r}{a+b\,x}+\frac{d\,q\,r}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0\wedge m\in\mathbb{Z}^+$, then

$$\int \frac{\text{Log}\left[i\,(j\,(h\,x)^t)^u\right]^m \text{Log}\left[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r\right]}{x}\,dx \longrightarrow$$

$$\frac{\text{Log}\left[i\,(j\,(h\,x)^t)^u\right]^{m+1} \text{Log}\left[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r\right]}{t\,u\,(m+1)}-\frac{b\,p\,r}{t\,u\,(m+1)}\int \frac{\text{Log}\left[i\,(j\,(h\,x)^t)^u\right]^{m+1}}{a+b\,x}\,dx-\frac{d\,q\,r}{t\,u\,(m+1)}\int \frac{\text{Log}\left[i\,(j\,(h\,x)^t)^u\right]^{m+1}}{c+d\,x}\,dx$$

Program code:

```
Int[Log[i_.*(j_.*(h_.*x_)^t_.)^u_.]^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]/x_,x_Symbol] :=
  Log[i*(j*(h*x)^t)^u]^(m+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]/(t*u*(m+1)) -
  b*p*r/(t*u*(m+1))*Int[Log[i*(j*(h*x)^t)^u]^(m+1)/(a+b*x),x] -
  d*q*r/(t*u*(m+1))*Int[Log[i*(j*(h*x)^t)^u]^(m+1)/(c+d*x),x] /;
FreeQ[{a,b,c,d,e,f,h,i,j,m,p,q,r,t,u},x] && NeQ[b*c-a*d,0] && IGtQ[m,0]
```

U: $\int \frac{\text{Log}\left[i\,(j\,(h\,x)^t)^u\right]^m \text{Log}\left[e\,(f\,(a+b\,x)^p\,(c+d\,x)^q)^r\right]^s}{x}\,dx$ when $b\,c-a\,d\neq 0$

Rule: If $b\,c-a\,d\neq 0$, then

$$\int \frac{\text{Log}\left[i\left(j\,(h\,x)^t\right)^u\right]^m\,\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{x}\,dx \longrightarrow \int \frac{\text{Log}\left[i\left(j\,(h\,x)^t\right)^u\right]^m\,\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{x}\,dx$$

Program code:

```
Int[Log[i_.*(j_.*(h_.*x_)^t_.)^u_.]^m_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_./x_,x_Symbol] :=
  Unintegrable[Log[i*(j*(h*x)^t)^u]^m*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/x,x] /;
FreeQ[{a,b,c,d,e,f,h,i,j,m,p,q,r,s,t,u},x] && NeQ[b*c-a*d,0]
```

5. $\int \frac{\text{Log}\left[e\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(g+h\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,s\in\mathbb{Z}^+\wedge\,p+q==0$

1: $\int \frac{\text{Log}\left[e\,\frac{c+d\,x}{a+b\,x}\right]}{(a+b\,x)\,(g+h\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,g\,(b-d\,e)-h\,(a-c\,e)==0$

Derivation: Integration by substitution

Basis: If $g\,(b-d\,e)-h\,(a-c\,e)==0$, then $\frac{\text{Log}\left[e\,\frac{c+d\,x}{a+b\,x}\right]}{(a+b\,x)\,(g+h\,x)}==-\frac{b-d\,e}{h\,(b\,c-a\,d)}\,\text{Subst}\left[\frac{\text{Log}[e\,x]}{1-e\,x},\,x,\,\frac{c+d\,x}{a+b\,x}\right]\partial_x\frac{c+d\,x}{a+b\,x}$

Rule: If $b\,c-a\,d\neq 0\,\wedge\,g\,(b-d\,e)-h\,(a-c\,e)==0$, then

$$\int \frac{\text{Log}\left[e\,\frac{c+d\,x}{a+b\,x}\right]}{(a+b\,x)\,(g+h\,x)}\,dx \longrightarrow -\frac{b-d\,e}{h\,(b\,c-a\,d)}\,\text{Subst}\left[\int\frac{\text{Log}[e\,x]}{1-e\,x}\,dx,\,x,\,\frac{c+d\,x}{a+b\,x}\right]$$

Program code:

```
Int[u_*Log[e_.*(c_.+d_.*x_)/(a_.+b_.*x_)],x_Symbol] :=
  With[{g=Coeff[Simplify[1/(u*(a+b*x))],x,0],h=Coeff[Simplify[1/(u*(a+b*x))],x,1]},
  -(b-d*e)/(h*(b*c-a*d))*Subst[Int[Log[e*x]/(1-e*x),x],x,(c+d*x)/(a+b*x)] /;
 EqQ[g*(b-d*e)-h*(a-c*e),0]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b*c-a*d,0] && LinearQ[Simplify[1/(u*(a+b*x))],x]
```

2: $\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(g+h\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\land\,s\in\mathbb{Z}^+\,\land\,p+q=0\,\land\,b\,g-a\,h\neq 0\,\land\,d\,g-c\,h\neq 0$

Derivation: Integration by parts

Basis: $\frac{1}{(a+b\,x)\,(g+h\,x)} == -\frac{1}{b\,g-a\,h}\,\partial_x\,\text{Log}\left[-\frac{(b\,c-a\,d)\,(g+h\,x)}{(d\,g-c\,h)\,(a+b\,x)}\right]$

Basis: If $p+q==0$, then $\partial_x\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s == \frac{p\,r\,s\,(b\,c-a\,d)}{(a+b\,x)\,(c+d\,x)}\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}$

Rule: If $b\,c-a\,d\neq 0\,\land\,s\in\mathbb{Z}^+\,\land\,p+q==0\,\land\,b\,g-a\,h\neq 0\,\land\,d\,g-c\,h\neq 0$, then

$$\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(g+h\,x)}\,dx \;\longrightarrow$$

$$-\frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{b\,g-a\,h}\,\text{Log}\left[-\frac{(b\,c-a\,d)\,(g+h\,x)}{(d\,g-c\,h)\,(a+b\,x)}\right]+$$
$$\frac{p\,r\,s\,(b\,c-a\,d)}{b\,g-a\,h}\int \frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{(a+b\,x)\,(c+d\,x)}\,\text{Log}\left[-\frac{(b\,c-a\,d)\,(g+h\,x)}{(d\,g-c\,h)\,(a+b\,x)}\right]dx$$

Program code:

```
Int[u_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{g=Coeff[Simplify[1/(u*(a+b*x))],x,0],h=Coeff[Simplify[1/(u*(a+b*x))],x,1]},
  -Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(b*g-a*h)*Log[-(b*c-a*d)*(g+h*x)/((d*g-c*h)*(a+b*x))] +
  p*r*s*(b*c-a*d)/(b*g-a*h)*
    Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((a+b*x)*(c+d*x))*Log[-(b*c-a*d)*(g+h*x)/((d*g-c*h)*(a+b*x))],x] /;
 NeQ[b*g-a*h,0] && NeQ[d*g-c*h,0]] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && EqQ[p+q,0] && LinearQ[Simplify[1/(u*(a+b*x))],x]
```

6. $\int \frac{u\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(c+d\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\land\,p+q==0$

1. $\int \frac{\text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s}{(a+b x)(c+d x)}\,dx$ when $b c-a d\neq 0\land p+q==0$

   1: $\int \frac{1}{(a+b x)(c+d x)\,\text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]}\,dx$ when $b c-a d\neq 0\land p+q==0$

Rule: If $b c-a d\neq 0\ \land\ p+q==0$, then

$$\int \frac{1}{(a+b x)(c+d x)\,\text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]}\,dx\ \longrightarrow\ \frac{\text{Log}\left[\text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]\right]}{p\,r\,(b c-a d)}$$

Program code:

```
Int[u_/Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.],x_Symbol] :=
  With[{h=Simplify[u*(a+b*x)*(c+d*x)]},
  h*Log[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]]/(p*r*(b*c-a*d)) /;
 FreeQ[h,x]] /;
FreeQ[{a,b,c,d,e,f,p,q,r},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0]
```

2: $\int \frac{\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(a+b x) (c+d x)} \, dx$ when $b c-a d\neq 0 \land p+q==0 \land s\neq -1$

Rule: If $b c-a d\neq 0 \land p+q==0 \land s\neq -1$, then

$$\int \frac{\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(a+b x) (c+d x)} \, dx \to \frac{\text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s+1}}{p r (s+1) (b c-a d)}$$

Program code:

```
Int[u_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{h=Simplify[u*(a+b*x)*(c+d*x)]},
  h*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s+1)/(p*r*(s+1)*(b*c-a*d)) /;
 FreeQ[h,x]] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && NeQ[s,-1]
```

2: $\int \frac{\text{Log}\left[1+g \frac{a+b x}{c+d x}\right] \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(a+b x) (c+d x)} \, dx$ when $b c-a d\neq 0 \land s\in \mathbb{Z}^+ \land p+q==0$

Derivation: Integration by parts

Basis: $\frac{\text{Log}\left[1+g \frac{a+b x}{c+d x}\right]}{(a+b x) (c+d x)} == -\partial_x \frac{\text{PolyLog}\left[2,-g \frac{a+b x}{c+d x}\right]}{b c-a d}$

Basis: If $p+q==0$, then $\partial_x \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s == \frac{p r s (b c-a d)}{(a+b x) (c+d x)} \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1}$

Rule: If $b c-a d\neq 0 \land s\in \mathbb{Z}^+ \land p+q==0$, then

$$\int \frac{\text{Log}\left[1+g \frac{a+b x}{c+d x}\right] \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(a+b x) (c+d x)} \, dx \to$$

$$-\frac{\text{PolyLog}\left[2,-g\,\frac{a+b\,x}{c+d\,x}\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{b\,c-a\,d}+p\,r\,s\int\frac{\text{PolyLog}\left[2,-g\,\frac{a+b\,x}{c+d\,x}\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s-1}}{(a+b\,x)\,(c+d\,x)}\,dx$$

Program code:

```
Int[u_*Log[v_]*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{g=Simplify[(v-1)*(c+d*x)/(a+b*x)],h=Simplify[u*(a+b*x)*(c+d*x)]},
  -h*PolyLog[2,1-v]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(b*c-a*d) +
  h*p*r*s*Int[PolyLog[2,1-v]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((a+b*x)*(c+d*x)),x] /;
 FreeQ[{g,h},x]] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && EqQ[p+q,0]
```

3: $\int \frac{\text{Log}\left[i\,\left(j\,(g+h\,x)^t\right)^u\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(c+d\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\land\,p+q=0\,\land\,s\neq -1$

Derivation: Integration by parts

Basis: If $p+q=0$, then $\frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(c+d\,x)}=\partial_x\frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s+1}}{p\,r\,(s+1)\,(b\,c-a\,d)}$

Basis: $\partial_x\,\text{Log}\left[i\,\left(j\,(g+h\,x)^t\right)^u\right]=\frac{h\,t\,u}{g+h\,x}$

Rule: If $b\,c-a\,d\neq 0\,\land\,p+q=0\,\land\,s\neq -1$, then

$$\int \frac{\text{Log}\left[i\,\left(j\,(g+h\,x)^t\right)^u\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(c+d\,x)}\,dx\,\longrightarrow$$

$$\frac{\text{Log}\left[i\,\left(j\,(g+h\,x)^t\right)^u\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s+1}}{p\,r\,(s+1)\,(b\,c-a\,d)}-\frac{h\,t\,u}{p\,r\,(s+1)\,(b\,c-a\,d)}\int\frac{\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^{s+1}}{g+h\,x}\,dx$$

Program code:

```
Int[v_*Log[i_.*(j_.*(g_.+h_.*x_)^t_.)^u_.]*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{k=Simplify[v*(a+b*x)*(c+d*x)]},
  k*Log[i*(j*(g+h*x)^t)^u]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s+1)/(p*r*(s+1)*(b*c-a*d)) -
  k*h*t*u/(p*r*(s+1)*(b*c-a*d))*Int[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s+1)/(g+h*x),x] /;
 FreeQ[k,x]] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,p,q,r,s,t,u},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && NeQ[s,-1]
```

4: $\int \frac{\text{PolyLog}\left[n,\,g\,\frac{a+b\,x}{c+d\,x}\right]\,\text{Log}\left[e\,\left(f\,(a+b\,x)^p\,(c+d\,x)^q\right)^r\right]^s}{(a+b\,x)\,(c+d\,x)}\,dx$ when $b\,c-a\,d\neq 0\,\land\,s\in\mathbb{Z}^+\land\,p+q=0$

Derivation: Integration by parts

Basis: $\frac{\text{PolyLog}\left[n,g\frac{a+b x}{c+d x}\right]}{(a+b x)(c+d x)} = \partial_x \frac{\text{PolyLog}\left[n+1,g\frac{a+b x}{c+d x}\right]}{b c-a d}$

Basis: If $p+q = 0$, then $\partial_x \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s = \frac{p r s (b c-a d)}{(a+b x)(c+d x)} \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^{s-1}$

Rule: If $b c-a d \neq 0 \wedge s \in \mathbb{Z}^+ \wedge p+q = 0$, then

$$\int \frac{\text{PolyLog}\left[n, g\frac{a+b x}{c+d x}\right] \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s}{(a+b x)(c+d x)} dx \to$$
$$\frac{\text{PolyLog}\left[n+1, g\frac{a+b x}{c+d x}\right] \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s}{b c-a d} - p r s \int \frac{\text{PolyLog}\left[n+1, g\frac{a+b x}{c+d x}\right] \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^{s-1}}{(a+b x)(c+d x)} dx$$

Program code:

```
Int[u_*PolyLog[n_,v_]*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{g=Simplify[v*(c+d*x)/(a+b*x)],h=Simplify[u*(a+b*x)*(c+d*x)]},
  h*PolyLog[n+1,v]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/(b*c-a*d) -
  h*p*r*s*Int[PolyLog[n+1,v]*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1)/((a+b*x)*(c+d*x)),x] /;
 FreeQ[{g,h},x]] /;
FreeQ[{a,b,c,d,e,f,n,p,q,r,s},x] && NeQ[b*c-a*d,0] && IGtQ[s,0] && EqQ[p+q,0]
```

7. $\int (a+b x)^m (c+d x)^n \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s dx$ when $b c-a d \neq 0 \wedge p+q = 0 \wedge m+n+2 = 0 \wedge m \neq -1$

1: $\int (a+b x)^m (c+d x)^n \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s dx$ when $b c-a d \neq 0 \wedge p+q = 0 \wedge m+n+2 = 0 \wedge m \neq -1 \wedge s \in \mathbb{Z}^+$

Derivation: Integration by parts

Basis: If $m+n+2 = 0$, then $(a+b x)^m (c+d x)^n = \partial_x \frac{(a+b x)^{m+1} (c+d x)^{n+1}}{(m+1)(b c-a d)}$

Basis: If $p+q = 0$, then $\partial_x \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^s = \frac{p r s (b c-a d)}{(a+b x)(c+d x)} \text{Log}\left[e\left(f(a+b x)^p (c+d x)^q\right)^r\right]^{s-1}$

Rule: If $b c-a d \neq 0 \wedge p+q = 0 \wedge m+n+2 = 0 \wedge m \neq -1 \wedge s \in \mathbb{Z}^+$, then

$$\int (a+b x)^m (c+d x)^n \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx \longrightarrow$$
$$\frac{(a+b x)^{m+1} (c+d x)^{n+1} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s}{(m+1) (b c-a d)}-\frac{p r s (b c-a d)}{(m+1) (b c-a d)}\int (a+b x)^m (c+d x)^n \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^{s-1} dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^(n+1)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s/((m+1)*(b*c-a*d)) -
  p*r*s*(b*c-a*d)/((m+1)*(b*c-a*d))*Int[(a+b*x)^m*(c+d*x)^n*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^(s-1),x] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r,s},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && EqQ[m+n+2,0] && NeQ[m,-1] && IGtQ[s,0]
```

2: $\int \frac{(a+b x)^m (c+d x)^n}{\operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]} dx$ when $b c-a d \neq 0 \land p+q == 0 \land m+n+2 == 0 \land m \neq -1$

Rule: If $b c-a d \neq 0 \land p+q == 0 \land m+n+2 == 0 \land m \neq -1$, then

$$\int \frac{(a+b x)^m (c+d x)^n}{\operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]} dx \longrightarrow$$

$$\frac{(a+b x)^{m+1} (c+d x)^{n+1}}{p r (b c-a d) \left(e \left(f (a+b x)^p (c+d x)^q\right)^r\right)^{\frac{m+1}{p r}}} \operatorname{ExpIntegralEi}\left[\frac{m+1}{p r} \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]\right]$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_./Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.],x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^(n+1)/(p*r*(b*c-a*d)*(e*(f*(a+b*x)^p*(c+d*x)^q)^r)^((m+1)/(p*r)))*
    ExpIntegralEi[(m+1)/(p*r)*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]] /;
FreeQ[{a,b,c,d,e,f,m,n,p,q,r},x] && NeQ[b*c-a*d,0] && EqQ[p+q,0] && EqQ[m+n+2,0] && NeQ[m,-1]
```

8: $\int \frac{\left(a+b\,\text{Log}\left[c\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]\right)^n}{A+B\,x+C\,x^2}\,dx$ when $C\,d\,f-A\,e\,g=0\,\wedge\,B\,e\,g-C\,(e\,f+d\,g)=0\,\wedge\,n\in\mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $F[x] = 2\,(e\,f-d\,g)\,\text{Subst}\left[\frac{x}{(e-g\,x^2)^2}\,F\left[-\frac{d-f\,x^2}{e-g\,x^2}\right],\,x,\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]\,\partial_x\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}$

Basis: If $C\,d\,f-A\,e\,g=0\,\wedge\,B\,e\,g-C\,(e\,f+d\,g)=0$, then
$\frac{1}{A+B\,x+C\,x^2} = \frac{2\,e\,g}{C\,(e\,f-d\,g)}\,\text{Subst}\left[\frac{1}{x},\,x,\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]\,\partial_x\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}$

Rule: If $C\,d\,f-A\,e\,g=0\,\wedge\,B\,e\,g-C\,(e\,f+d\,g)=0\,\wedge\,n\in\mathbb{Z}^+$, then

$$\int \frac{\left(a+b\,\text{Log}\left[c\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]\right)^n}{A+B\,x+C\,x^2}\,dx \to \frac{2\,e\,g}{C\,(e\,f-d\,g)}\,\text{Subst}\left[\int\frac{(a+b\,\text{Log}[c\,x])^n}{x}\,dx,\,x,\,\frac{\sqrt{d+e\,x}}{\sqrt{f+g\,x}}\right]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*Sqrt[d_.+e_.*x_]/Sqrt[f_.+g_.*x_]])^n_./(A_.+B_.*x_+C_.*x_^2),x_Symbol] :=
  2*e*g/(C*(e*f-d*g))*Subst[Int[(a+b*Log[c*x])^n/x,x],x,Sqrt[d+e*x]/Sqrt[f+g*x]] /;
FreeQ[{a,b,c,d,e,f,g,A,B,C,n},x] && EqQ[C*d*f-A*e*g,0] && EqQ[B*e*g-C*(e*f+d*g),0]
```

```
Int[(a_.+b_.*Log[c_.*Sqrt[d_.+e_.*x_]/Sqrt[f_.+g_.*x_]])^n_./(A_.+C_.*x_^2),x_Symbol] :=
  g/(C*f)*Subst[Int[(a+b*Log[c*x])^n/x,x],x,Sqrt[d+e*x]/Sqrt[f+g*x]] /;
FreeQ[{a,b,c,d,e,f,g,A,C,n},x] && EqQ[C*d*f-A*e*g,0] && EqQ[e*f+d*g,0]
```

9. $\int RF_x \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx$

1: $\int RF_x \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] dx$ when $b c - a d \neq 0$

Derivation: Algebraic expansion and piecewise constant extraction

Basis: $u A == u B + u C - (B + C - A) u$

Basis: $\partial_x \left(p r \text{Log}[a+b x] + q r \text{Log}[c+d x] - \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]\right) == 0$

Rule: If $b c - a d \neq 0$, then

$$\int RF_x \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right] dx \longrightarrow$$

$$p r \int RF_x \text{Log}[a+b x] dx + q r \int RF_x \text{Log}[c+d x] dx - \left(p r \text{Log}[a+b x] + q r \text{Log}[c+d x] - \text{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]\right) \int RF_x dx$$

Program code:

```
Int[RFx_.*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.],x_Symbol] :=
  p*r*Int[RFx*Log[a+b*x],x] +
  q*r*Int[RFx*Log[c+d*x],x] -
  (p*r*Log[a+b*x]+q*r*Log[c+d*x] - Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r])*Int[RFx,x] /;
FreeQ[{a,b,c,d,e,f,p,q,r},x] && RationalFunctionQ[RFx,x] && NeQ[b*c-a*d,0] &&
  Not[MatchQ[RFx,u_.*(a+b*x)^m_.*(c+d*x)^n_. /; IntegersQ[m,n]]]
```

x: $\int RF_x \,\mathrm{Log}\left[e\left(f(a+b\,x)^p(c+d\,x)^q\right)^r\right] dx$ when $b\,c-a\,d\neq 0$

Derivation: Integration by parts

Basis: $\partial_x \mathrm{Log}\left[e\left(f(a+b\,x)^p(c+d\,x)^q\right)^r\right] = \frac{b\,p\,r}{a+b\,x}+\frac{d\,q\,r}{c+d\,x}$

Rule: If $b\,c-a\,d\neq 0$, let $u\to\int RF_x\,dx$, then

$$\int RF_x \,\mathrm{Log}\left[e\left(f(a+b\,x)^p(c+d\,x)^q\right)^r\right] dx \longrightarrow u\,\mathrm{Log}\left[e\left(f(a+b\,x)^p(c+d\,x)^q\right)^r\right]-b\,p\,r\int\frac{u}{a+b\,x}dx-d\,q\,r\int\frac{u}{c+d\,x}dx$$

Program code:

```
(* Int[RFx_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.],x_Symbol] :=
  With[{u=IntHide[RFx,x]},
  u*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r] - b*p*r*Int[u/(a+b*x),x] - d*q*r*Int[u/(c+d*x),x] /;
 NonsumQ[u]] /;
FreeQ[{a,b,c,d,e,f,p,q,r},x] && RationalFunctionQ[RFx,x] && NeQ[b*c-a*d,0] *)
```

2: $\int RF_x \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx$ when $s \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $s \in \mathbb{Z}^+$, then

$$\int RF_x \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx \rightarrow \int \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s \operatorname{ExpandIntegrand}[RF_x, x]\, dx$$

Program code:

```
Int[RFx_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  With[{u=ExpandIntegrand[Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s,RFx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && RationalFunctionQ[RFx,x] && IGtQ[s,0]
```

U: $\int RF_x \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx$

Rule:

$$\int RF_x \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx \rightarrow \int RF_x \operatorname{Log}\left[e \left(f (a+b x)^p (c+d x)^q\right)^r\right]^s dx$$

Program code:

```
Int[RFx_*Log[e_.*(f_.*(a_.+b_.*x_)^p_.*(c_.+d_.*x_)^q_.)^r_.]^s_.,x_Symbol] :=
  Unintegrable[RFx*Log[e*(f*(a+b*x)^p*(c+d*x)^q)^r]^s,x] /;
FreeQ[{a,b,c,d,e,f,p,q,r,s},x] && RationalFunctionQ[RFx,x]
```

N: $\int u \, \text{Log}\left[e \left(f \, v^p \, w^q\right)^r\right]^s dx$ when $v == a + b x \land w == c + d x$

Derivation: Algebraic normalization

Rule: If $v == a + b x \land w == c + d x$, then

$$\int u \, \text{Log}\left[e \left(f \, v^p \, w^q\right)^r\right]^s dx \longrightarrow \int u \, \text{Log}\left[e \left(f \, (a + b x)^p \, (c + d x)^q\right)^r\right]^s dx$$

Program code:

```
Int[u_.*Log[e_.*(f_.*v_^p_.*w_^q_.)^r_.]^s_.,x_Symbol] :=
  Int[u*Log[e*(f*ExpandToSum[v,x]^p*ExpandToSum[w,x]^q)^r]^s,x] /;
FreeQ[{e,f,p,q,r,s},x] && LinearQ[{v,w},x] && Not[LinearMatchQ[{v,w},x]] && AlgebraicFunctionQ[u,x]
```

```
Int[u_.*Log[e_.*(f_.*(g_+v_./w_))^r_.]^s_.,x_Symbol] :=
  Int[u*Log[e*(f*ExpandToSum[v+g*w,x]/ExpandToSum[w,x])^r]^s,x] /;
FreeQ[{e,f,g,r,s},x] && LinearQ[w,x] && (FreeQ[v,x] || LinearQ[v,x]) && AlgebraicFunctionQ[u,x]
```

x: $\int \frac{\text{Log}\left[i\,\left(j\,\left(g+h\,x\right)^s\right)^t\right]\,\text{Log}\left[e\,\left(f\,\left(a+b\,x\right)^p\,\left(c+d\,x\right)^q\right)^r\right]}{m+n\,x}\,dx$

Derivation: Integration by substitution

Basis: $F[x] = \frac{1}{n}\,\text{Subst}\left[F\left[\frac{x-m}{n}\right],\,x,\,m+n\,x\right]\,\partial_x\,(m+n\,x)$

Rule:

$$\int \frac{\text{Log}\left[i\,\left(j\,\left(g+h\,x\right)^s\right)^t\right]\,\text{Log}\left[e\,\left(f\,\left(a+b\,x\right)^p\,\left(c+d\,x\right)^q\right)^r\right]}{m+n\,x}\,dx\,\longrightarrow$$
$$\frac{1}{n}\,\text{Subst}\left[\int \frac{1}{x}\text{Log}\left[i\,\left(j\,\left(-\frac{h\,m-g\,n}{n}+\frac{h\,x}{n}\right)^s\right)^t\right]\,\text{Log}\left[e\,\left(f\,\left(-\frac{b\,m-a\,n}{n}+\frac{b\,x}{n}\right)^p\,\left(-\frac{d\,m-c\,n}{n}+\frac{d\,x}{n}\right)^q\right)^r\right]\,dx,\,x,\,m+n\,x\right]$$

Program code:

```
(* Int[Log[g_.*(h_.*(a_.+b_.*x_)^p_.)^q_.]*Log[i_.*(j_.*(c_.+d_.*x_)^r_.)^s_.]/(e_+f_.*x_),x_Symbol] :=
  1/f*Subst[Int[Log[g*(h*Simp[-(b*e-a*f)/f+b*x/f,x]^p)^q]*Log[i*(j*Simp[-(d*e-c*f)/f+d*x/f,x]^r)^s]/x,x],x,e+f*x] /;
FreeQ[{a,b,c,d,e,f,g,h,i,j,p,q,r,s},x] *)
```