

Rules for integrands of the form $\text{Trig}[c + d x]^m (a \cos[c + d x] + b \sin[c + d x])^n$

1. $\int (a \cos[c + d x] + b \sin[c + d x])^n dx$

1: $\int (a \cos[c + d x] + b \sin[c + d x])^n dx$ when $a^2 + b^2 = 0$

Reference: Integration by substitution

Basis: If $a^2 + b^2 = 0$, then

$$(a \cos[c + d x] + b \sin[c + d x])^n = \frac{a (a \cos[c + d x] + b \sin[c + d x])^{n-1}}{b d} \partial_x (a \cos[c + d x] + b \sin[c + d x])$$

Rule: If $a^2 + b^2 = 0$, then

$$\int (a \cos[c + d x] + b \sin[c + d x])^n dx \rightarrow \frac{a (a \cos[c + d x] + b \sin[c + d x])^n}{b d n}$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  a*(a*cos[c+d*x]+b*sin[c+d*x])^n/(b*d*n) /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2+b^2,0]
```

$$2. \int (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0$$

$$1. \int (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge n > 1$$

$$1: \int (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge \frac{n-1}{2} \in \mathbb{Z}^+$$

Reference: G&R 2.557'

Derivation: Integration by substitution

Basis: If $\frac{n-1}{2} \in \mathbb{Z}$, then

$$(a \cos[z] + b \sin[z])^n = - \left(a^2 + b^2 - (b \cos[z] - a \sin[z])^2 \right)^{\frac{n-1}{2}} \partial_z (b \cos[z] - a \sin[z])$$

Rule: If $a^2 + b^2 \neq 0 \wedge \frac{n-1}{2} \in \mathbb{Z}^+$, then

$$\int (a \cos[c+dx] + b \sin[c+dx])^n dx \rightarrow -\frac{1}{d} \text{Subst} \left[\int (a^2 + b^2 - x^2)^{\frac{n-1}{2}} dx, x, b \cos[c+dx] - a \sin[c+dx] \right]$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -1/d*Subst[Int[(a^2+b^2-x^2)^(n-1)/2,x],x,b*Cos[c+d*x]-a*Sin[c+d*x]] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && IGtQ[(n-1)/2,0]
```

$$2: \int (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge \frac{n-1}{2} \notin \mathbb{Z} \wedge n > 1$$

Derivation: Integration by parts with a double-back flip

Rule: If $a^2 + b^2 \neq 0 \wedge \frac{n-1}{2} \notin \mathbb{Z} \wedge n > 1$, then

$$\int (a \cos[c+dx] + b \sin[c+dx])^n dx \rightarrow$$

$$- \frac{(b \cos[c+dx] - a \sin[c+dx]) (a \cos[c+dx] + b \sin[c+dx])^{n-1}}{dn} + \frac{(n-1)(a^2+b^2)}{n} \int (a \cos[c+dx] + b \sin[c+dx])^{n-2} dx$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -(b*cos[c+d*x]-a*sin[c+d*x])*(a*cos[c+d*x]+b*sin[c+d*x])^(n-1)/(d*n) +
  (n-1)*(a^2+b^2)/n*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && Not[IntegerQ[(n-1)/2]] && GtQ[n,1]
```

2. $\int (a \cos[c+dx] + b \sin[c+dx])^n dx$ when $a^2 + b^2 \neq 0 \wedge n \leq -1$

1: $\int \frac{1}{a \cos[c+dx] + b \sin[c+dx]} dx$ when $a^2 + b^2 \neq 0$

Reference: G&R 2.557'

Derivation: Integration by substitution

Basis: If $\frac{n-1}{2} \in \mathbb{Z}$, then

$$(a \cos[z] + b \sin[z])^n = - (a^2 + b^2 - (b \cos[z] - a \sin[z])^2)^{\frac{n-1}{2}} \partial_z (b \cos[z] - a \sin[z])$$

Rule: If $a^2 + b^2 \neq 0$, then

$$\int \frac{1}{a \cos[c+dx] + b \sin[c+dx]} dx \rightarrow -\frac{1}{d} \text{Subst}\left[\int \frac{1}{a^2 + b^2 - x^2} dx, x, b \cos[c+dx] - a \sin[c+dx]\right]$$

Program code:

```
Int[1/(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  -1/d*Subst[Int[1/(a^2+b^2-x^2),x],x,b*cos[c+d*x]-a*sin[c+d*x]] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

$$\text{2: } \int \frac{1}{(a \cos[c+dx] + b \sin[c+dx])^2} dx \text{ when } a^2 + b^2 \neq 0$$

Reference: G&R 2.557.5b'

Rule: If $a^2 + b^2 \neq 0$, then

$$\int \frac{1}{(a \cos[c+dx] + b \sin[c+dx])^2} dx \rightarrow \frac{\sin[c+dx]}{ad (a \cos[c+dx] + b \sin[c+dx])}$$

Program code:

```
Int[1/(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^2,x_Symbol] :=
  Sin[c+d*x]/(a*d*(a*cos[c+d*x]+b*sin[c+d*x])) /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

$$\text{3: } \int (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge n < -1 \wedge n \neq -2$$

Derivation: Integration by parts with a double-back flip

Rule: If $a^2 + b^2 \neq 0 \wedge n < -1 \wedge n \neq -2$, then

$$\int (a \cos[c+dx] + b \sin[c+dx])^n dx \rightarrow \frac{(b \cos[c+dx] - a \sin[c+dx]) (a \cos[c+dx] + b \sin[c+dx])^{n+1}}{d(n+1)(a^2+b^2)} + \frac{n+2}{(n+1)(a^2+b^2)} \int (a \cos[c+dx] + b \sin[c+dx])^{n+2} dx$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  (b*cos[c+d*x]-a*sin[c+d*x])*(a*cos[c+d*x]+b*sin[c+d*x])^(n+1)/(d*(n+1)*(a^2+b^2)) +
  (n+2)/((n+1)*(a^2+b^2))*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n+2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1] && NeQ[n,-2]
```

$$3. \int (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge \neg (n \geq 1 \vee n \leq -1)$$

$$1: \int (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } \neg (n \geq 1 \vee n \leq -1) \wedge a^2 + b^2 > 0$$

Derivation: Algebraic simplification

$$\text{Basis: If } a^2 + b^2 \neq 0, \text{ then } a \cos[z] + b \sin[z] = \sqrt{a^2 + b^2} \cos[z - \text{ArcTan}[a, b]]$$

Rule: If $\neg (n \geq 1 \vee n \leq -1) \wedge a^2 + b^2 > 0$, then

$$\int (a \cos[c+dx] + b \sin[c+dx])^n dx \rightarrow (a^2 + b^2)^{n/2} \int (\cos[c+dx - \text{ArcTan}[a, b]])^n dx$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  (a^2+b^2)^(n/2)*Int[(Cos[c+d*x-ArcTan[a,b]])^n,x] /;
FreeQ[{a,b,c,d,n},x] && Not[GeQ[n,1] || LeQ[n,-1]] && GtQ[a^2+b^2,0]
```

$$2: \int (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } \neg (n \geq 1 \vee n \leq -1) \wedge \neg (a^2 + b^2 \geq 0)$$

Derivation: Piecewise constant extraction and algebraic simplification

$$\text{Basis: } \partial_x \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\left(\frac{a \cos[c+dx] + b \sin[c+dx]}{\sqrt{a^2 + b^2}} \right)^n} = 0$$

$$\text{Basis: If } a^2 + b^2 \neq 0, \text{ then } \frac{a \cos[z] + b \sin[z]}{\sqrt{a^2 + b^2}} = \cos[z - \text{ArcTan}[a, b]]$$

Rule: If $\neg (n \geq 1 \vee n \leq -1) \wedge \neg (a^2 + b^2 \geq 0)$, then

$$\int (a \cos[c+dx] + b \sin[c+dx])^n dx \rightarrow \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\left(\frac{a \cos[c+dx] + b \sin[c+dx]}{\sqrt{a^2+b^2}}\right)^n} \int \left(\frac{a \cos[c+dx] + b \sin[c+dx]}{\sqrt{a^2+b^2}}\right)^n dx$$

$$\rightarrow \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\left(\frac{a \cos[c+dx] + b \sin[c+dx]}{\sqrt{a^2+b^2}}\right)^n} \int (\cos[c+dx - \text{ArcTan}[a, b]])^n dx$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  (a*Cos[c+d*x]+b*Sin[c+d*x])^n/((a*Cos[c+d*x]+b*Sin[c+d*x])/Sqrt[a^2+b^2])^n*Int[Cos[c+d*x-ArcTan[a,b]]^n,x] /;
FreeQ[{a,b,c,d,n},x] && Not[GeQ[n,1] || LeQ[n,-1]] && Not[GtQ[a^2+b^2,0] || EqQ[a^2+b^2,0]]
```

$$2. \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx$$

$$1. \int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]^n} dx \text{ when } n \in \mathbb{Z}$$

$$1. \int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]^n} dx \text{ when } n \in \mathbb{Z} \wedge a^2 + b^2 = 0$$

$$1: \int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]^n} dx \text{ when } a^2 + b^2 = 0 \wedge n > 1$$

Note: Compare this with the rule for integrands of the form $(a + b \cot[c+dx])^n$ when $a^2 + b^2 = 0 \wedge n > 1$.

Rule: If $a^2 + b^2 = 0 \wedge n > 1$, then

$$\int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]^n} dx \rightarrow -\frac{a (a \cos[c+dx] + b \sin[c+dx])^{n-1}}{d (n-1) \sin[c+dx]^{n-1}} + 2b \int \frac{(a \cos[c+dx] + b \sin[c+dx])^{n-1}}{\sin[c+dx]^{n-1}} dx$$

Program code:

```
Int[sin[c_+d_.*x_]^m_*(a_.*cos[c_+d_.*x_]+b_.*sin[c_+d_.*x_])^n_,x_Symbol] :=
  -a*(a*cos[c+d*x]+b*sin[c+d*x])^(n-1)/(d*(n-1)*sin[c+d*x]^(n-1)) +
  2*b*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n-1)/sin[c+d*x]^(n-1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[m+n,0] && EqQ[a^2+b^2,0] && GtQ[n,1]
```

```
Int[cos[c_+d_.*x_]^m_*(a_.*cos[c_+d_.*x_]+b_.*sin[c_+d_.*x_])^n_,x_Symbol] :=
  b*(a*cos[c+d*x]+b*sin[c+d*x])^(n-1)/(d*(n-1)*cos[c+d*x]^(n-1)) +
  2*a*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n-1)/cos[c+d*x]^(n-1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[m+n,0] && EqQ[a^2+b^2,0] && GtQ[n,1]
```

$$2: \int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]^n} dx \text{ when } a^2 + b^2 = 0 \wedge n < 0$$

Note: Compare this with the rule for integrands of the form $(a + b \cot[c+dx])^n$ when $a^2 + b^2 = 0 \wedge n < 0$.

Rule: If $a^2 + b^2 = 0 \wedge n < 0$, then

$$\int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]^n} dx \rightarrow \frac{a (a \cos[c+dx] + b \sin[c+dx])^n}{2 b d n \sin[c+dx]^n} + \frac{1}{2 b} \int \frac{(a \cos[c+dx] + b \sin[c+dx])^{n+1}}{\sin[c+dx]^{n+1}} dx$$

Program code:

```
Int[sin[c_.+d_.**x_]^m_.*(a_.*cos[c_.+d_.**x_]+b_.*sin[c_.+d_.**x_])^n_,x_Symbol] :=
  a*(a*cos[c+d*x]+b*sin[c+d*x])^n/(2*b*d*n*sin[c+d*x]^n) +
  1/(2*b)*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n+1)/sin[c+d*x]^(n+1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[m+n,0] && EqQ[a^2+b^2,0] && LtQ[n,0]
```

```
Int[cos[c_.+d_.**x_]^m_.*(a_.*cos[c_.+d_.**x_]+b_.*sin[c_.+d_.**x_])^n_,x_Symbol] :=
  -b*(a*cos[c+d*x]+b*sin[c+d*x])^n/(2*a*d*n*cos[c+d*x]^n) +
  1/(2*a)*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n+1)/cos[c+d*x]^(n+1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[m+n,0] && EqQ[a^2+b^2,0] && LtQ[n,0]
```


$$\text{3: } \int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]^n} dx \text{ when } a^2 + b^2 = 0 \wedge n \notin \mathbb{Z}$$

Rule: If $a^2 + b^2 = 0 \wedge n \notin \mathbb{Z}$, then

$$\int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]^n} dx \rightarrow \frac{a (a \cos[c+dx] + b \sin[c+dx])^n}{2 b d n \sin[c+dx]^n} \text{Hypergeometric2F1}\left[1, n, n+1, \frac{b + a \cot[c+dx]}{2 b}\right]$$

Program code:

```
Int[sin[c_.+d_.**x_]^m_.*(a_.*cos[c_.+d_.**x_]+b_.*sin[c_.+d_.**x_])^n_,x_Symbol] :=
  a*(a*Cos[c+d*x]+b*Sin[c+d*x])^n/(2*b*d*n*Sin[c+d*x]^n)*Hypergeometric2F1[1,n,n+1,(b+a*Cot[c+d*x])/(2*b)] /;
FreeQ[{a,b,c,d,n},x] && EqQ[m+n,0] && EqQ[a^2+b^2,0] && Not[IntegerQ[n]]
```

```
Int[cos[c_.+d_.**x_]^m_.*(a_.*cos[c_.+d_.**x_]+b_.*sin[c_.+d_.**x_])^n_,x_Symbol] :=
  -b*(a*Cos[c+d*x]+b*Sin[c+d*x])^n/(2*a*d*n*Cos[c+d*x]^n)*Hypergeometric2F1[1,n,n+1,(a+b*Tan[c+d*x])/(2*a)] /;
FreeQ[{a,b,c,d,n},x] && EqQ[m+n,0] && EqQ[a^2+b^2,0] && Not[IntegerQ[n]]
```

$$2: \int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]^n} dx \text{ when } n \in \mathbb{Z} \wedge a^2 + b^2 \neq 0$$

Derivation: Algebraic simplification

$$\text{Basis: } \frac{a \cos[z] + b \sin[z]}{\sin[z]} = b + a \cot[z]$$

Rule: If $n \in \mathbb{Z} \wedge a^2 + b^2 \neq 0$, then

$$\int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]^n} dx \rightarrow \int (b + a \cot[c+dx])^n dx$$

Program code:

```
Int[sin[c_+d_.*x_]^m_*(a_.*cos[c_+d_.*x_]+b_.*sin[c_+d_.*x_])^n_,x_Symbol] :=
  Int[(b+a*Cot[c+d*x])^n,x] /;
  FreeQ[{a,b,c,d},x] && EqQ[m+n,0] && IntegerQ[n] && NeQ[a^2+b^2,0]
```

```
Int[cos[c_+d_.*x_]^m_*(a_.*cos[c_+d_.*x_]+b_.*sin[c_+d_.*x_])^n_,x_Symbol] :=
  Int[(a+b*Tan[c+d*x])^n,x] /;
  FreeQ[{a,b,c,d},x] && EqQ[m+n,0] && IntegerQ[n] && NeQ[a^2+b^2,0]
```

$$2: \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } n \in \mathbb{Z} \wedge \frac{m+n}{2} \in \mathbb{Z}$$

Derivation: Integration by substitution

$$\text{Basis: If } n \in \mathbb{Z}, \text{ then } \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n = \sin[c+dx]^{m+n} \frac{(a+b \tan[c+dx])^n}{\tan[c+dx]^n}$$

$$\text{Basis: If } \frac{m+n}{2} \in \mathbb{Z}, \text{ then } \sin[c+dx]^{m+n} \frac{(a+b \tan[c+dx])^n}{\tan[c+dx]^n} = \frac{1}{d} \frac{\tan[c+dx]^m (a+b \tan[c+dx])^n}{(1+\tan[c+dx]^2)^{\frac{m+n+2}{2}}} \partial_x \tan[c+dx]$$

Rule: If $n \in \mathbb{Z} \wedge \frac{m+n}{2} \in \mathbb{Z}$, then

$$\int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \rightarrow \frac{1}{d} \text{Subst}\left[\int \frac{x^m (a+bx)^n}{(1+x^2)^{\frac{m+n+2}{2}}} dx, x, \tan[c+dx]\right]$$

Program code:

```
Int[sin[c_+d_*x_]^m_.*(a_*cos[c_+d_*x_]+b_*sin[c_+d_*x_])^n_,x_Symbol] :=
  1/d*Subst[Int[x^m*(a+b*x)^n/(1+x^2)^((m+n+2)/2),x],x,Tan[c+d*x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[n] && IntegerQ[(m+n)/2] && NeQ[n,-1] && Not[GtQ[n,0]] && GtQ[m,1]
```

```
Int[cos[c_+d_*x_]^m_.*(a_*cos[c_+d_*x_]+b_*sin[c_+d_*x_])^n_,x_Symbol] :=
  -1/d*Subst[Int[x^m*(b+a*x)^n/(1+x^2)^((m+n+2)/2),x],x,Cot[c+d*x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[n] && IntegerQ[(m+n)/2] && NeQ[n,-1] && Not[GtQ[n,0]] && GtQ[m,1]
```

3: $\int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx$ when $m \in \mathbb{Z} \wedge n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $m \in \mathbb{Z} \wedge n \in \mathbb{Z}^+$, then

$$\int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \rightarrow \int \text{ExpandTrig}[\sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n, x] dx$$

Program code:

```
Int[sin[c_+d_*x_]^m_.*(a_*cos[c_+d_*x_]+b_*sin[c_+d_*x_])^n_,x_Symbol] :=
  Int[ExpandTrig[sin[c+d*x]^m*(a*cos[c+d*x]+b*sin[c+d*x])^n,x],x] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && IGtQ[n,0]
```

```
Int[cos[c_+d_*x_]^m_.*(a_*cos[c_+d_*x_]+b_*sin[c_+d_*x_])^n_,x_Symbol] :=
  Int[ExpandTrig[cos[c+d*x]^m*(a*cos[c+d*x]+b*sin[c+d*x])^n,x],x] /;
FreeQ[{a,b,c,d},x] && IntegerQ[m] && IGtQ[n,0]
```

4: $\int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx$ when $a^2 + b^2 = 0 \wedge n \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: If $a^2 + b^2 = 0$, then $a \cos[z] + b \sin[z] = ab (b \cos[z] + a \sin[z])^{-1}$

Rule: If $a^2 + b^2 = 0 \wedge n \in \mathbb{Z}^-$, then

$$\int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \rightarrow a^n b^n \int \sin[c+dx]^m (b \cos[c+dx] + a \sin[c+dx])^{-n} dx$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  a^n*b^n*Int[Sin[c+d*x]^m*(b*cos[c+d*x]+a*sin[c+d*x])^(-n),x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[a^2+b^2,0] && ILtQ[n,0]
```

```
Int[cos[c_.+d_.*x_]^m_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  a^n*b^n*Int[Cos[c+d*x]^m*(b*cos[c+d*x]+a*sin[c+d*x])^(-n),x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[a^2+b^2,0] && ILtQ[n,0]
```

$$5. \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0$$

$$1. \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge n > 0$$

$$2. \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge n > 1$$

$$1. \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge n > 1 \wedge m > 0$$

$$2. \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge n > 1 \wedge m < 0$$

$$1: \int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]} dx \text{ when } a^2 + b^2 \neq 0 \wedge n > 1$$

Derivation: Algebraic expansion and power rule for integration

$$\text{Basis: } \frac{(a \cos[z] + b \sin[z])^2}{\sin[z]} = a (b \cos[z] - a \sin[z]) + b (a \cos[z] + b \sin[z]) + \frac{a^2}{\sin[z]}$$

Rule: If $a^2 + b^2 \neq 0 \wedge n < -1$, then

$$\int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]} dx \rightarrow \frac{a (a \cos[c+dx] + b \sin[c+dx])^{n-1}}{d (n-1)} + b \int (a \cos[c+dx] + b \sin[c+dx])^{n-1} dx + a^2 \int \frac{(a \cos[c+dx] + b \sin[c+dx])^{n-2}}{\sin[c+dx]} dx$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_/sin[c_.+d_.*x_],x_Symbol] :=
  a*(a*cos[c+d*x]+b*sin[c+d*x])^(n-1)/(d*(n-1)) +
  b*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n-1),x] +
  a^2*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n-2)/sin[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1]
```

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_/cos[c_.+d_.*x_],x_Symbol] :=
  -b*(a*cos[c+d*x]+b*sin[c+d*x])^(n-1)/(d*(n-1)) +
  a*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n-1),x] +
  b^2*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n-2)/cos[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1]
```

2: $\int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx$ when $a^2 + b^2 \neq 0 \wedge n > 1 \wedge m < -1$

Derivation: Algebraic expansion

Basis: $(a \cos[z] + b \sin[z])^2 = -(a^2 + b^2) \sin[z]^2 + 2b \sin[z] (a \cos[z] + b \sin[z]) + a^2$

Rule: If $a^2 + b^2 \neq 0 \wedge n > 1 \wedge m < -1$, then

$$\begin{aligned} & \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \rightarrow \\ & - (a^2 + b^2) \int \sin[c+dx]^{m+2} (a \cos[c+dx] + b \sin[c+dx])^{n-2} dx + \\ & 2b \int \sin[c+dx]^{m+1} (a \cos[c+dx] + b \sin[c+dx])^{n-1} dx + a^2 \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^{n-2} dx \end{aligned}$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -(a^2+b^2)*Int[Sin[c+d*x]^(m+2)*(a*cos[c+d*x]+b*sin[c+d*x])^(n-2),x] +
  2*b*Int[Sin[c+d*x]^(m+1)*(a*cos[c+d*x]+b*sin[c+d*x])^(n-1),x] +
  a^2*Int[Sin[c+d*x]^m*(a*cos[c+d*x]+b*sin[c+d*x])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && GtQ[n,1] && LtQ[m,-1]
```

```
Int[cos[c_.+d_.*x_]^m_*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_,x_Symbol] :=
  -(a^2+b^2)*Int[Cos[c+d*x]^(m+2)*(a*cos[c+d*x]+b*sin[c+d*x])^(n-2),x] +
  2*a*Int[Cos[c+d*x]^(m+1)*(a*cos[c+d*x]+b*sin[c+d*x])^(n-1),x] +
  b^2*Int[Cos[c+d*x]^m*(a*cos[c+d*x]+b*sin[c+d*x])^(n-2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && GtQ[n,1] && LtQ[m,-1]
```

$$2. \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge n < 0$$

$$1. \int \frac{\sin[c+dx]^m}{a \cos[c+dx] + b \sin[c+dx]} dx \text{ when } a^2 + b^2 \neq 0$$

$$1. \int \frac{\sin[c+dx]^m}{a \cos[c+dx] + b \sin[c+dx]} dx \text{ when } a^2 + b^2 \neq 0 \wedge m > 0$$

$$1: \int \frac{\sin[c+dx]}{a \cos[c+dx] + b \sin[c+dx]} dx \text{ when } a^2 + b^2 \neq 0$$

Derivation: Algebraic expansion

$$\text{Basis: } \frac{\sin[z]}{a \cos[z] + b \sin[z]} == \frac{b}{a^2 + b^2} - \frac{a(b \cos[z] - a \sin[z])}{(a^2 + b^2)(a \cos[z] + b \sin[z])}$$

Rule: If $a^2 + b^2 \neq 0$, then

$$\int \frac{\sin[c+dx]}{a \cos[c+dx] + b \sin[c+dx]} dx \rightarrow \frac{bx}{a^2 + b^2} - \frac{a}{a^2 + b^2} \int \frac{b \cos[c+dx] - a \sin[c+dx]}{a \cos[c+dx] + b \sin[c+dx]} dx$$

Program code:

```
Int[sin[c_+d_.x_]/(a_.cos[c_+d_.x_]+b_.sin[c_+d_.x_]),x_Symbol] :=
  b*x/(a^2+b^2) -
  a/(a^2+b^2)*Int[(b*cos[c+d*x]-a*sin[c+d*x])/(a*cos[c+d*x]+b*sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

```
Int[cos[c_+d_.x_]/(a_.cos[c_+d_.x_]+b_.sin[c_+d_.x_]),x_Symbol] :=
  a*x/(a^2+b^2) +
  b/(a^2+b^2)*Int[(b*cos[c+d*x]-a*sin[c+d*x])/(a*cos[c+d*x]+b*sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

$$\mathbf{2:} \int \frac{\sin[c+dx]^m}{a \cos[c+dx] + b \sin[c+dx]} dx \text{ when } a^2 + b^2 \neq 0 \wedge m > 1$$

Derivation: Algebraic expansion and power rule for integration

$$\text{Basis: } \frac{\sin[z]^2}{a \cos[z] + b \sin[z]} == -\frac{a \cos[z]}{a^2 + b^2} + \frac{b \sin[z]}{a^2 + b^2} + \frac{a^2}{(a^2 + b^2)(a \cos[z] + b \sin[z])}$$

Rule: If $a^2 + b^2 \neq 0 \wedge m > 1$, then

$$\int \frac{\sin[c+dx]^m}{a \cos[c+dx] + b \sin[c+dx]} dx \rightarrow -\frac{a \sin[c+dx]^{m-1}}{d(a^2 + b^2)(m-1)} + \frac{b}{a^2 + b^2} \int \sin[c+dx]^{m-1} dx + \frac{a^2}{a^2 + b^2} \int \frac{\sin[c+dx]^{m-2}}{a \cos[c+dx] + b \sin[c+dx]} dx$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_/ (a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]), x_Symbol] :=
  -a*Sin[c+d*x]^(m-1)/(d*(a^2+b^2)*(m-1)) +
  b/(a^2+b^2)*Int[Sin[c+d*x]^(m-1), x] +
  a^2/(a^2+b^2)*Int[Sin[c+d*x]^(m-2)/(a*cos[c+d*x]+b*sin[c+d*x]), x] /;
FreeQ[{a,b,c,d}, x] && NeQ[a^2+b^2, 0] && GtQ[m, 1]
```

```
Int[cos[c_.+d_.*x_]^m_/ (a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]), x_Symbol] :=
  b*cos[c+d*x]^(m-1)/(d*(a^2+b^2)*(m-1)) +
  a/(a^2+b^2)*Int[Cos[c+d*x]^(m-1), x] +
  b^2/(a^2+b^2)*Int[Cos[c+d*x]^(m-2)/(a*cos[c+d*x]+b*sin[c+d*x]), x] /;
FreeQ[{a,b,c,d}, x] && NeQ[a^2+b^2, 0] && GtQ[m, 1]
```


$$2. \int \frac{\sin[c+dx]^m}{a \cos[c+dx] + b \sin[c+dx]} dx \text{ when } a^2 + b^2 \neq 0 \wedge m < 0$$

$$1: \int \frac{1}{\sin[c+dx] (a \cos[c+dx] + b \sin[c+dx])} dx \text{ when } a^2 + b^2 \neq 0$$

Derivation: Algebraic expansion

$$\text{Basis: } \frac{1}{\sin[z] (a \cos[z] + b \sin[z])} == \frac{\cot[z]}{a} - \frac{b \cos[z] - a \sin[z]}{a (a \cos[z] + b \sin[z])}$$

Rule: If $a^2 + b^2 \neq 0$, then

$$\int \frac{1}{\sin[c+dx] (a \cos[c+dx] + b \sin[c+dx])} dx \rightarrow \frac{1}{a} \int \cot[c+dx] dx - \frac{1}{a} \int \frac{b \cos[c+dx] - a \sin[c+dx]}{a \cos[c+dx] + b \sin[c+dx]} dx$$

Program code:

```
Int[1/(sin[c_.+d_.*x_]*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])),x_Symbol] :=
  1/a*Int[Cot[c+d*x],x] -
  1/a*Int[(b*cos[c+d*x]-a*sin[c+d*x])/(a*cos[c+d*x]+b*sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

```
Int[1/(cos[c_.+d_.*x_]*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])),x_Symbol] :=
  1/b*Int[Tan[c+d*x],x] +
  1/b*Int[(b*cos[c+d*x]-a*sin[c+d*x])/(a*cos[c+d*x]+b*sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0]
```

$$2: \int \frac{\sin[c+dx]^m}{a \cos[c+dx] + b \sin[c+dx]} dx \text{ when } a^2 + b^2 \neq 0 \wedge m < -1$$

Derivation: Algebraic expansion and power rule for integration

$$\text{Basis: } \frac{1}{a \cos[z] + b \sin[z]} == \frac{\cos[z]}{a} - \frac{b \sin[z]}{a^2} + \frac{(a^2+b^2) \sin[z]^2}{a^2 (a \cos[z] + b \sin[z])}$$

Rule: If $a^2 + b^2 \neq 0 \wedge m < -1$, then

$$\int \frac{\sin[c+dx]^m}{a \cos[c+dx] + b \sin[c+dx]} dx \rightarrow \frac{\sin[c+dx]^{m+1}}{a d (m+1)} - \frac{b}{a^2} \int \sin[c+dx]^{m+1} dx + \frac{a^2 + b^2}{a^2} \int \frac{\sin[c+dx]^{m+2}}{a \cos[c+dx] + b \sin[c+dx]} dx$$

Program code:

```
Int[sin[c_.+d_.*x_]^m_/(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  Sin[c+d*x]^(m+1)/(a*d*(m+1)) -
  b/a^2*Int[Sin[c+d*x]^(m+1),x] +
  (a^2+b^2)/a^2*Int[Sin[c+d*x]^(m+2)/(a*cos[c+d*x]+b*sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[m,-1]
```

```
Int[cos[c_.+d_.*x_]^m_/(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  -Cos[c+d*x]^(m+1)/(b*d*(m+1)) -
  a/b^2*Int[Cos[c+d*x]^(m+1),x] +
  (a^2+b^2)/b^2*Int[Cos[c+d*x]^(m+2)/(a*cos[c+d*x]+b*sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[m,-1]
```

$$2. \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge n < -1$$

$$1. \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge n < -1 \wedge m > 0$$

$$2. \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge n < -1 \wedge m < 0$$

$$1: \int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]} dx \text{ when } a^2 + b^2 \neq 0 \wedge n < -1$$

Derivation: Algebraic expansion and power rule for integration

$$\text{Basis: } \frac{1}{\sin[z]} == -\frac{(b \cos[z] - a \sin[z])}{a} - \frac{b(a \cos[z] + b \sin[z])}{a^2} + \frac{(a \cos[z] + b \sin[z])^2}{a^2 \sin[z]}$$

Rule: If $a^2 + b^2 \neq 0 \wedge n < -1$, then

$$\int \frac{(a \cos[c+dx] + b \sin[c+dx])^n}{\sin[c+dx]} dx \rightarrow$$

$$-\frac{(a \cos[c+dx] + b \sin[c+dx])^{n+1}}{ad(n+1)} - \frac{b}{a^2} \int (a \cos[c+dx] + b \sin[c+dx])^{n+1} dx + \frac{1}{a^2} \int \frac{(a \cos[c+dx] + b \sin[c+dx])^{n+2}}{\sin[c+dx]} dx$$

Program code:

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_/sin[c_.+d_.*x_],x_Symbol] :=
  -(a*cos[c+d*x]+b*sin[c+d*x])^(n+1)/(a*d*(n+1)) -
  b/a^2*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n+1),x] +
  1/a^2*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n+2)/sin[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1]
```

```
Int[(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^n_/cos[c_.+d_.*x_],x_Symbol] :=
  (a*cos[c+d*x]+b*sin[c+d*x])^(n+1)/(b*d*(n+1)) -
  a/b^2*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n+1),x] +
  1/b^2*Int[(a*cos[c+d*x]+b*sin[c+d*x])^(n+2)/cos[c+d*x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1]
```

$$\mathbf{2:} \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \text{ when } a^2 + b^2 \neq 0 \wedge n < -1 \wedge m < -1$$

Derivation: Algebraic expansion

$$\text{Basis: } 1 == \frac{(a^2+b^2) \sin[z]^2}{a^2} - \frac{2 b \sin[z] (a \cos[z]+b \sin[z])}{a^2} + \frac{(a \cos[z]+b \sin[z])^2}{a^2}$$

Rule: If $a^2 + b^2 \neq 0 \wedge n < -1 \wedge m < -1$, then

$$\begin{aligned} & \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^n dx \rightarrow \\ & \frac{a^2 + b^2}{a^2} \int \sin[c+dx]^{m+2} (a \cos[c+dx] + b \sin[c+dx])^n dx - \\ & \frac{2b}{a^2} \int \sin[c+dx]^{m+1} (a \cos[c+dx] + b \sin[c+dx])^{n+1} dx + \frac{1}{a^2} \int \sin[c+dx]^m (a \cos[c+dx] + b \sin[c+dx])^{n+2} dx \end{aligned}$$

Program code:

```
Int[sin[c_+d_*x_]^m_*(a_*cos[c_+d_*x_]+b_*sin[c_+d_*x_])^n_,x_Symbol] :=
  (a^2+b^2)/a^2*Int[Sin[c+d*x]^(m+2)*(a*cos[c+d*x]+b*sin[c+d*x])^n,x] -
  2*b/a^2*Int[Sin[c+d*x]^(m+1)*(a*cos[c+d*x]+b*sin[c+d*x])^(n+1),x] +
  1/a^2*Int[Sin[c+d*x]^m*(a*cos[c+d*x]+b*sin[c+d*x])^(n+2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1] && LtQ[m,-1]
```

```
Int[cos[c_+d_*x_]^m_*(a_*cos[c_+d_*x_]+b_*sin[c_+d_*x_])^n_,x_Symbol] :=
  (a^2+b^2)/b^2*Int[Cos[c+d*x]^(m+2)*(a*cos[c+d*x]+b*sin[c+d*x])^n,x] -
  2*a/b^2*Int[Cos[c+d*x]^(m+1)*(a*cos[c+d*x]+b*sin[c+d*x])^(n+1),x] +
  1/b^2*Int[Cos[c+d*x]^m*(a*cos[c+d*x]+b*sin[c+d*x])^(n+2),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && LtQ[n,-1] && LtQ[m,-1]
```

$$3. \int \cos[c+dx]^m \sin[c+dx]^n (a \cos[c+dx] + b \sin[c+dx])^p dx$$

$$1. \int \cos[c+dx]^m \sin[c+dx]^n (a \cos[c+dx] + b \sin[c+dx])^p dx \text{ when } p > 0$$

$$\mathbf{1:} \int \cos[c+dx]^m \sin[c+dx]^n (a \cos[c+dx] + b \sin[c+dx])^p dx \text{ when } p \in \mathbb{Z}^+$$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int \cos[c+dx]^m \sin[c+dx]^n (a \cos[c+dx] + b \sin[c+dx])^p dx \rightarrow \int \text{ExpandTrig}[\cos[c+dx]^m \sin[c+dx]^n (a \cos[c+dx] + b \sin[c+dx])^p, x] dx$$

Program code:

```
Int[cos[c_+d_.x_]^m_.sin[c_+d_.x_]^n_.(a_.cos[c_+d_.x_]+b_.sin[c_+d_.x_])^p_,x_Symbol] :=
  Int[ExpandTrig[cos[c+d*x]^m*sin[c+d*x]^n*(a*cos[c+d*x]+b*sin[c+d*x])^p,x],x] /;
FreeQ[{a,b,c,d,m,n},x] && IGtQ[p,0]
```

$$2. \int \cos[c+dx]^m \sin[c+dx]^n (a \cos[c+dx] + b \sin[c+dx])^p dx \text{ when } p < 0$$

$$1: \int \cos[c+dx]^m \sin[c+dx]^n (a \cos[c+dx] + b \sin[c+dx])^p dx \text{ when } a^2 + b^2 \neq 0 \wedge p \in \mathbb{Z}^-$$

Derivation: Algebraic simplification

Basis: If $a^2 + b^2 \neq 0$, then $a \cos[z] + b \sin[z] = \frac{1}{\sqrt{a^2 + b^2}} (b \cos[z] + a \sin[z])^{-1}$

Rule: If $a^2 + b^2 \neq 0 \wedge p \in \mathbb{Z}^-$, then

$$\int \cos[c+dx]^m \sin[c+dx]^n (a \cos[c+dx] + b \sin[c+dx])^p dx \rightarrow a^p b^p \int \cos[c+dx]^m \sin[c+dx]^n (b \cos[c+dx] + a \sin[c+dx])^{-p} dx$$

Program code:

```
Int[cos[c_.+d_.*x_]^m_.*sin[c_.+d_.*x_]^n_.*(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_])^p_,x_Symbol] :=
  a^p*b^p*Int[Cos[c+d*x]^m*Sin[c+d*x]^n*(b*Cos[c+d*x]+a*Sin[c+d*x])^(-p),x] /;
  FreeQ[{a,b,c,d,m,n},x] && EqQ[a^2+b^2,0] && ILtQ[p,0]
```

$$2. \int \frac{\cos[c+dx]^m \sin[c+dx]^n}{a \cos[c+dx] + b \sin[c+dx]} dx$$

$$1: \int \frac{\cos[c+dx]^m \sin[c+dx]^n}{a \cos[c+dx] + b \sin[c+dx]} dx \text{ when } a^2 + b^2 \neq 0 \wedge m \in \mathbb{Z}^+ \wedge n \in \mathbb{Z}^+$$

Derivation: Algebraic expansion

$$\text{Basis: } \frac{\cos[z] \sin[z]}{a \cos[z] + b \sin[z]} = \frac{b \cos[z]}{a^2 + b^2} + \frac{a \sin[z]}{a^2 + b^2} - \frac{a b}{(a^2 + b^2) (a \cos[z] + b \sin[z])}$$

Rule: If $a^2 + b^2 \neq 0 \wedge m \in \mathbb{Z}^+ \wedge n \in \mathbb{Z}^+$, then

$$\int \frac{\cos[c+dx]^m \sin[c+dx]^n}{a \cos[c+dx] + b \sin[c+dx]} dx \rightarrow$$

$$\frac{b}{a^2 + b^2} \int \cos[c+dx]^m \sin[c+dx]^{n-1} dx + \frac{a}{a^2 + b^2} \int \cos[c+dx]^{m-1} \sin[c+dx]^n dx - \frac{ab}{a^2 + b^2} \int \frac{\cos[c+dx]^{m-1} \sin[c+dx]^{n-1}}{a \cos[c+dx] + b \sin[c+dx]} dx$$

Program code:

```
Int[cos[c_.+d_.*x_]^m_.*sin[c_.+d_.*x_]^n_./(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  b/(a^2+b^2)*Int[Cos[c+d*x]^m*Sin[c+d*x]^(n-1),x] +
  a/(a^2+b^2)*Int[Cos[c+d*x]^(m-1)*Sin[c+d*x]^n,x] -
  a*b/(a^2+b^2)*Int[Cos[c+d*x]^(m-1)*Sin[c+d*x]^(n-1)/(a*cos[c+d*x]+b*sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && IGtQ[m,0] && IGtQ[n,0]
```

2: $\int \frac{\cos[c+dx]^m \sin[c+dx]^n}{a \cos[c+dx] + b \sin[c+dx]} dx$ when $(m | n) \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $(m | n) \in \mathbb{Z}$, then

$$\int \frac{\cos[c+dx]^m \sin[c+dx]^n}{a \cos[c+dx] + b \sin[c+dx]} dx \rightarrow \int \text{ExpandTrig}\left[\frac{\cos[c+dx]^m \sin[c+dx]^n}{a \cos[c+dx] + b \sin[c+dx]}, x\right] dx$$

Program code:

```
Int[cos[c_.+d_.*x_]^m_.*sin[c_.+d_.*x_]^n_./(a_.*cos[c_.+d_.*x_]+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  Int[ExpandTrig[cos[c+d*x]^m*sin[c+d*x]^n/(a*cos[c+d*x]+b*sin[c+d*x]),x],x] /;
FreeQ[{a,b,c,d,m,n},x] && IntegersQ[m,n]
```

$$\mathbf{3:} \int \cos[c+dx]^m \sin[c+dx]^n (a \cos[c+dx] + b \sin[c+dx])^p dx \text{ when } a^2 + b^2 \neq 0 \wedge m \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}^+ \wedge p \in \mathbb{Z}^-$$

Derivation: Algebraic expansion

$$\text{Basis: } \frac{\cos[z] \sin[z]}{a \cos[z] + b \sin[z]} == \frac{b \cos[z]}{a^2 + b^2} + \frac{a \sin[z]}{a^2 + b^2} - \frac{a b}{(a^2 + b^2) (a \cos[z] + b \sin[z])}$$

Rule: If $a^2 + b^2 \neq 0 \wedge m \in \mathbb{Z}^+ \wedge m \in \mathbb{Z}^+ \wedge p \in \mathbb{Z}^-$, then

$$\begin{aligned} & \int \cos[c+dx]^m \sin[c+dx]^n (a \cos[c+dx] + b \sin[c+dx])^p dx \rightarrow \\ & \frac{b}{a^2 + b^2} \int \cos[c+dx]^m \sin[c+dx]^{n-1} (a \cos[c+dx] + b \sin[c+dx])^{p+1} dx + \\ & \frac{a}{a^2 + b^2} \int \cos[c+dx]^{m-1} \sin[c+dx]^n (a \cos[c+dx] + b \sin[c+dx])^{p+1} dx - \\ & \frac{a b}{a^2 + b^2} \int \cos[c+dx]^{m-1} \sin[c+dx]^{n-1} (a \cos[c+dx] + b \sin[c+dx])^p dx \end{aligned}$$

Program code:

```
Int[cos[c_+d_.x_]^m_.sin[c_+d_.x_]^n_.*(a_.cos[c_+d_.x_]+b_.sin[c_+d_.x_])^p_,x_Symbol] :=
  b/(a^2+b^2)*Int[Cos[c+d*x]^m*Sin[c+d*x]^(n-1)*(a*cos[c+d*x]+b*sin[c+d*x])^(p+1),x] +
  a/(a^2+b^2)*Int[Cos[c+d*x]^(m-1)*Sin[c+d*x]^n*(a*cos[c+d*x]+b*sin[c+d*x])^(p+1),x] -
  a*b/(a^2+b^2)*Int[Cos[c+d*x]^(m-1)*Sin[c+d*x]^(n-1)*(a*cos[c+d*x]+b*sin[c+d*x])^p,x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2+b^2,0] && IGtQ[m,0] && IGtQ[n,0] && ILtQ[p,0]
```