

## Rules for integrands of the form $(g \cos[e + f x])^p (a + b \sin[e + f x])^m$

1.  $\int \cos[e + f x]^p (a + b \sin[e + f x])^m dx$  when  $\frac{p-1}{2} \in \mathbb{Z}$

**1:**  $\int \cos[e + f x]^p (a + b \sin[e + f x])^m dx$  when  $\frac{p-1}{2} \in \mathbb{Z} \wedge a^2 - b^2 = 0$

Derivation: Integration by substitution

Basis: If  $\frac{p-1}{2} \in \mathbb{Z} \wedge a^2 - b^2 = 0$ , then

$$\cos[e + f x]^p (a + b \sin[e + f x])^m = \frac{1}{b^p f} \text{Subst}\left[(a + x)^{m + \frac{p-1}{2}} (a - x)^{\frac{p-1}{2}}, x, b \sin[e + f x]\right] \partial_x (b \sin[e + f x])$$

Rule: If  $\frac{p-1}{2} \in \mathbb{Z} \wedge a^2 - b^2 = 0$ , then

$$\int \cos[e + f x]^p (a + b \sin[e + f x])^m dx \rightarrow \frac{1}{b^p f} \text{Subst}\left[\int (a + x)^{m + \frac{p-1}{2}} (a - x)^{\frac{p-1}{2}} dx, x, b \sin[e + f x]\right]$$

Program code:

```
Int[cos[e_+f_.*x_]^p_.*(a_+b_.*sin[e_+f_.*x_])^m_,x_Symbol] :=
  1/(b^p*f)*Subst[Int[(a+x)^(m+(p-1)/2)*(a-x)^((p-1)/2),x],x,b*Sin[e+f*x]] /;
FreeQ[{a,b,e,f,m},x] && IntegerQ[(p-1)/2] && EqQ[a^2-b^2,0] && (GeQ[p,-1] || Not[IntegerQ[m+1/2]])
```

**2:**  $\int \cos[e+fx]^p (a+b \sin[e+fx])^m dx$  when  $\frac{p-1}{2} \in \mathbb{Z} \wedge a^2 - b^2 \neq 0$

Derivation: Integration by substitution

Basis: If  $\frac{p-1}{2} \in \mathbb{Z}$ , then  $\cos[e+fx]^p F[b \sin[e+fx]] = \frac{1}{b^p f} \text{Subst}[F[x] (b^2 - x^2)^{\frac{p-1}{2}}, x, b \sin[e+fx]] \partial_x (b \sin[e+fx])$

Rule: If  $\frac{p-1}{2} \in \mathbb{Z} \wedge a^2 - b^2 \neq 0$ , then

$$\int \cos[e+fx]^p (a+b \sin[e+fx])^m dx \rightarrow \frac{1}{b^p f} \text{Subst}\left[\int (a+x)^m (b^2 - x^2)^{\frac{p-1}{2}} dx, x, b \sin[e+fx]\right]$$

Program code:

```
Int[cos[e_+f_.*x_]^p_.*(a_+b_.*sin[e_+f_.*x_])^m_,x_Symbol] :=
  1/(b^p*f)*Subst[Int[(a+x)^m*(b^2-x^2)^(p-1)/2],x,x,b*Sin[e+f*x]] /;
FreeQ[{a,b,e,f,m},x] && IntegerQ[(p-1)/2] && NeQ[a^2-b^2,0]
```

**2:**  $\int (g \cos[e+fx])^p (a+b \sin[e+fx]) dx$

Derivation: Nondegenerate sine recurrence 1b with  $c \rightarrow 0$ ,  $d \rightarrow 1$ ,  $A \rightarrow 0$ ,  $B \rightarrow a$ ,  $C \rightarrow b$ ,  $m \rightarrow 0$ ,  $n \rightarrow -1$

Rule:

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx]) dx \rightarrow -\frac{b (g \cos[e+fx])^{p+1}}{f g (p+1)} + a \int (g \cos[e+fx])^p dx$$

Program code:

```
Int[(g_.*cos[e_+f_.*x_])^p_.*(a_+b_.*sin[e_+f_.*x_]),x_Symbol] :=
  -b*(g*cos[e+f*x])^(p+1)/(f*g*(p+1)) + a*Int[(g*cos[e+f*x])^p,x] /;
FreeQ[{a,b,e,f,g,p},x] && (IntegerQ[2*p] || NeQ[a^2-b^2,0])
```

3.  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 = 0$

1:  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 = 0 \wedge m \in \mathbb{Z} \wedge p < -1 \wedge 2m+p \geq 0$

Derivation: Algebraic simplification

Basis: If  $a^2 - b^2 = 0 \wedge m \in \mathbb{Z}$ , then  $(a+b \sin[z])^m = \frac{a^{2m} \cos[z]^{2m}}{(a-b \sin[z])^m}$

Note: This rule removes removable singularities from the integrand and hence from the resulting antiderivatives.

Rule: If  $a^2 - b^2 = 0 \wedge m \in \mathbb{Z} \wedge p < -1 \wedge 2m+p \geq 0$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \frac{a^{2m}}{g^{2m}} \int \frac{(g \cos[e+fx])^{2m+p}}{(a-b \sin[e+fx])^m} dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  (a/g)^(2*m)*Int[(g*cos[e+f*x])^(2*m+p)/(a-b*sin[e+f*x])^m,x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && IntegerQ[m] && LtQ[p,-1] && GeQ[2*m+p,0]
```

2.  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 = 0 \wedge m+p \in \mathbb{Z}^-$

1:  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 = 0 \wedge m+p+1 = 0 \wedge p \notin \mathbb{Z}^-$

Derivation: Symmetric cosine/sine recurrence 1b with  $m \rightarrow -m-1$

Derivation: Symmetric cosine/sine recurrence 2c with  $m \rightarrow -m-1$

Rule: If  $a^2 - b^2 = 0 \wedge m+p+1 = 0 \wedge p \notin \mathbb{Z}^-$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \frac{b (g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^m}{a f g m}$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  b*(g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^m/(a*f*g*m) /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && EqQ[Simplify[m+p+1],0] && Not[ILtQ[p,0]]
```

**2:**  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 = 0 \wedge m+p+1 \in \mathbb{Z}^- \wedge 2m+p+1 \neq 0$

Derivation: Symmetric cosine/sine recurrence 2c

Rule: If  $a^2 - b^2 = 0 \wedge m+p+1 \in \mathbb{Z}^- \wedge 2m+p+1 \neq 0$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \frac{b (g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^m}{a f g (2m+p+1)} + \frac{m+p+1}{a (2m+p+1)} \int (g \cos[e+fx])^p (a+b \sin[e+fx])^{m+1} dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  b*(g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^m/(a*f*g*Simplify[2*m+p+1]) +
  Simplify[m+p+1]/(a*Simplify[2*m+p+1])*Int[(g*cos[e+f*x])^p*(a+b*sin[e+f*x])^(m+1),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && ILtQ[Simplify[m+p+1],0] && NeQ[2*m+p+1,0] && Not[IGtQ[m,0]]
```

$$3. \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge \frac{2m+p+1}{2} \in \mathbb{Z}^+$$

$$1: \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge 2m+p-1 = 0 \wedge m \neq 1$$

Derivation: Symmetric cosine/sine recurrence 1a with  $m \rightarrow -2m+1$

Derivation: Symmetric cosine/sine recurrence 1c with  $m \rightarrow -2m+1$

Rule: If  $a^2 - b^2 = 0 \wedge 2m+p-1 = 0 \wedge m \neq 1$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \frac{b (g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^{m-1}}{f g (m-1)}$$

Program code:

```
Int[(g_.cos[e_.+f_.*x_])^p_*(a_+b_.sin[e_.+f_.*x_])^m_,x_Symbol] :=
  b*(g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^(m-1)/(f*g*(m-1)) /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && EqQ[2*m+p-1,0] && NeQ[m,1]
```

**2:**  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 = 0 \wedge \frac{2m+p-1}{2} \in \mathbb{Z}^+ \wedge m+p \neq 0$

Derivation: Symmetric cosine/sine recurrence 1c

Rule: If  $a^2 - b^2 = 0 \wedge \frac{2m+p-1}{2} \in \mathbb{Z}^+ \wedge m+p \neq 0$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow$$

$$- \frac{b (g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^{m-1}}{f g (m+p)} + \frac{a (2m+p-1)}{m+p} \int (g \cos[e+fx])^p (a+b \sin[e+fx])^{m-1} dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a_+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  -b*(g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^(m-1)/(f*g*(m+p)) +
  a*(2*m+p-1)/(m+p)*Int[(g*cos[e+f*x])^p*(a+b*sin[e+f*x])^(m-1),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && IGtQ[Simplify[(2*m+p-1)/2],0] && NeQ[m+p,0]
```

$$4. \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge m > 0$$

$$1. \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge m > 0 \wedge p < -1$$

$$1: \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge m > 0 \wedge p \leq -2m$$

Derivation: Symmetric cosine/sine recurrence 1b

Rule: If  $a^2 - b^2 = 0 \wedge m > 0 \wedge p \leq -2m$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow -\frac{b (g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^m}{a f g (p+1)} + \frac{a (m+p+1)}{g^2 (p+1)} \int (g \cos[e+fx])^{p+2} (a+b \sin[e+fx])^{m-1} dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  -b*(g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^m/(a*f*g*(p+1)) +
  a*(m+p+1)/(g^2*(p+1))*Int[(g*cos[e+f*x])^(p+2)*(a+b*sin[e+f*x])^(m-1),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && GtQ[m,0] && LeQ[p,-2*m] && IntegersQ[m+1/2,2*p]
```

$$\mathbf{2:} \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge m > 1 \wedge p < -1$$

Derivation: Symmetric cosine/sine recurrence 1a

Rule: If  $a^2 - b^2 = 0 \wedge m > 1 \wedge p < -1$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow -\frac{2b (g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^{m-1}}{f g (p+1)} + \frac{b^2 (2m+p-1)}{g^2 (p+1)} \int (g \cos[e+fx])^{p+2} (a+b \sin[e+fx])^{m-2} dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a_+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
-2*b*(g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^(m-1)/(f*g*(p+1)) +
b^2*(2*m+p-1)/(g^2*(p+1))*Int[(g*cos[e+f*x])^(p+2)*(a+b*sin[e+f*x])^(m-2),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && GtQ[m,1] && LtQ[p,-1] && IntegersQ[2*m,2*p]
```

$$\mathbf{2.} \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge m > 0 \wedge p \neq -1$$

$$\mathbf{1:} \int \frac{\sqrt{a+b \sin[e+fx]}}{\sqrt{g \cos[e+fx]}} dx \text{ when } a^2 - b^2 = 0$$

Derivation: Piecewise constant extraction and algebraic expansion

Basis: If  $a^2 - b^2 = 0$ , then  $\partial_x \frac{\sqrt{1+\cos[e+fx]} \sqrt{a+b \sin[e+fx]}}{a+a \cos[e+fx]+b \sin[e+fx]} = 0$

Rule: If  $a^2 - b^2 = 0$ , then

$$\int \frac{\sqrt{a+b \sin[e+fx]}}{\sqrt{g \cos[e+fx]}} dx \rightarrow \frac{\sqrt{1+\cos[e+fx]} \sqrt{a+b \sin[e+fx]}}{a+a \cos[e+fx]+b \sin[e+fx]} \int \frac{a+a \cos[e+fx]+b \sin[e+fx]}{\sqrt{g \cos[e+fx]} \sqrt{1+\cos[e+fx]}} dx$$



$$\rightarrow \frac{a \sqrt{1 + \cos[e + f x]} \sqrt{a + b \sin[e + f x]}}{a + a \cos[e + f x] + b \sin[e + f x]} \int \frac{\sqrt{1 + \cos[e + f x]}}{\sqrt{g \cos[e + f x]}} dx + \frac{b \sqrt{1 + \cos[e + f x]} \sqrt{a + b \sin[e + f x]}}{a + a \cos[e + f x] + b \sin[e + f x]} \int \frac{\sin[e + f x]}{\sqrt{g \cos[e + f x]} \sqrt{1 + \cos[e + f x]}} dx$$

Program code:

```
Int[Sqrt[a_+b_.sin[e_+f_.x_]]/Sqrt[g_.cos[e_+f_.x_]],x_Symbol] :=
  a*Sqrt[1+Cos[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]/(a+a*Cos[e+f*x]+b*Sin[e+f*x])*Int[Sqrt[1+Cos[e+f*x]]/Sqrt[g*Cos[e+f*x]],x] +
  b*Sqrt[1+Cos[e+f*x]]*Sqrt[a+b*Sin[e+f*x]]/(a+a*Cos[e+f*x]+b*Sin[e+f*x])*Int[Sin[e+f*x]/(Sqrt[g*Cos[e+f*x]]*Sqrt[1+Cos[e+f*x]]),x]
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0]
```

**2:**  $\int (g \cos[e + f x])^p (a + b \sin[e + f x])^m dx$  when  $a^2 - b^2 = 0 \wedge m > 0 \wedge m + p \neq 0$

Derivation: Symmetric cosine/sine recurrence 1c

Rule: If  $a^2 - b^2 = 0 \wedge m > 0 \wedge m + p \neq 0$ , then

$$\int (g \cos[e + f x])^p (a + b \sin[e + f x])^m dx \rightarrow$$

$$- \frac{b (g \cos[e + f x])^{p+1} (a + b \sin[e + f x])^{m-1}}{f g (m+p)} + \frac{a (2m+p-1)}{m+p} \int (g \cos[e + f x])^p (a + b \sin[e + f x])^{m-1} dx$$

Program code:

```
Int[(g_.cos[e_+f_.x_])^p_*(a_+b_.sin[e_+f_.x_])^m_,x_Symbol] :=
  -b*(g*Cos[e+f*x])^(p+1)*(a+b*Sin[e+f*x])^(m-1)/(f*g*(m+p)) +
  a*(2*m+p-1)/(m+p)*Int[(g*Cos[e+f*x])^p_*(a+b*Sin[e+f*x])^(m-1),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && GtQ[m,0] && NeQ[m+p,0] && IntegersQ[2*m,2*p]
```

$$5. \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge m < -1$$

$$1. \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge m < -1 \wedge p > 1$$

$$1: \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge m < -1 \wedge p > 1 \wedge (m > -2 \vee p + 2m + 1 = 0)$$

Derivation: Symmetric cosine/sine recurrence 2a and 1c

Rule: If  $a^2 - b^2 = 0 \wedge m < -1 \wedge p > 1 \wedge (m > -2 \vee p + 2m + 1 = 0)$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \frac{g (g \cos[e+fx])^{p-1} (a+b \sin[e+fx])^{m+1}}{b f (m+p)} + \frac{g^2 (p-1)}{a (m+p)} \int (g \cos[e+fx])^{p-2} (a+b \sin[e+fx])^{m+1} dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  g*(g*cos[e+f*x])^(p-1)*(a+b*sin[e+f*x])^(m+1)/(b*f*(m+p)) +
  g^2*(p-1)/(a*(m+p))*Int[(g*cos[e+f*x])^(p-2)*(a+b*sin[e+f*x])^(m+1),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && LtQ[m,-1] && GtQ[p,1] && (GtQ[m,-2] || EqQ[2*m+p+1,0] || EqQ[m,-2] && IntegerQ[p]) &&
  NeQ[m+p,0] && IntegerQ[2*m,2*p]
```

**2:**  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 = 0 \wedge m \leq -2 \wedge p > 1 \wedge 2m+p+1 \neq 0$

Derivation: Symmetric cosine/sine recurrence 2a

Rule: If  $a^2 - b^2 = 0 \wedge m \leq -2 \wedge p > 1 \wedge 2m+p+1 \neq 0$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \frac{2g (g \cos[e+fx])^{p-1} (a+b \sin[e+fx])^{m+1}}{b f (2m+p+1)} + \frac{g^2 (p-1)}{b^2 (2m+p+1)} \int (g \cos[e+fx])^{p-2} (a+b \sin[e+fx])^{m+2} dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  2*g*(g*cos[e+f*x])^(p-1)*(a+b*sin[e+f*x])^(m+1)/(b*f*(2*m+p+1)) +
  g^2*(p-1)/(b^2*(2*m+p+1))*Int[(g*cos[e+f*x])^(p-2)*(a+b*sin[e+f*x])^(m+2),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && LeQ[m,-2] && GtQ[p,1] && NeQ[2*m+p+1,0] && Not[IntegerQ[2*m,2*p]]
```

$$\mathbf{2:} \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge m < -1 \wedge 2m+p+1 \neq 0$$

Derivation: Symmetric cosine/sine recurrence 2c

Rule: If  $a^2 - b^2 = 0 \wedge m < -1 \wedge 2m+p+1 \neq 0$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \frac{b (g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^m}{a f g (2m+p+1)} + \frac{m+p+1}{a (2m+p+1)} \int (g \cos[e+fx])^p (a+b \sin[e+fx])^{m+1} dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  b*(g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^m/(a*f*g*(2*m+p+1)) +
  (m+p+1)/(a*(2*m+p+1))*Int[(g*cos[e+f*x])^p*(a+b*sin[e+f*x])^(m+1),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && LtQ[m,-1] && NeQ[2*m+p+1,0] && IntegersQ[2*m,2*p]
```

$$6. \int \frac{(g \cos[e+fx])^p}{a+b \sin[e+fx]} dx \text{ when } a^2 - b^2 = 0$$

$$1: \int \frac{(g \cos[e+fx])^p}{a+b \sin[e+fx]} dx \text{ when } a^2 - b^2 = 0 \wedge p > 1$$

Derivation: Symmetric cosine/sine recurrence 2a and 1c

Rule: If  $a^2 - b^2 = 0 \wedge p > 1$ , then

$$\int \frac{(g \cos[e+fx])^p}{a+b \sin[e+fx]} dx \rightarrow \frac{g (g \cos[e+fx])^{p-1}}{b f (p-1)} + \frac{g^2}{a} \int (g \cos[e+fx])^{p-2} dx$$

Program code:

```
Int[(g_.cos[e_.+f_.*x_])^p_/(a_+b_.sin[e_.+f_.*x_]),x_Symbol] :=
  g*(g*cos[e+f*x])^(p-1)/(b*f*(p-1)) + g^2/a*Int[(g*cos[e+f*x])^(p-2),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && GtQ[p,1] && IntegerQ[2*p]
```

**2:**  $\int \frac{(g \cos[e+fx])^p}{a+b \sin[e+fx]} dx$  when  $a^2 - b^2 = 0 \wedge p \neq 1$

Derivation: Symmetric cosine/sine recurrence 2c

Rule: If  $a^2 - b^2 = 0 \wedge p < 0$ , then

$$\int \frac{(g \cos[e+fx])^p}{a+b \sin[e+fx]} dx \rightarrow \frac{b (g \cos[e+fx])^{p+1}}{a f g (p-1) (a+b \sin[e+fx])} + \frac{p}{a (p-1)} \int (g \cos[e+fx])^p dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_/(a_+b_.sin[e_.+f_.x_]),x_Symbol] :=
  b*(g*cos[e+f*x])^(p+1)/(a*f*g*(p-1)*(a+b*sin[e+f*x])) +
  p/(a*(p-1))*Int[(g*cos[e+f*x])^p,x] /;
FreeQ[{a,b,e,f,g,p},x] && EqQ[a^2-b^2,0] && Not[GeQ[p,1]] && IntegerQ[2*p]
```

$$7. \int \frac{(g \cos[e+fx])^p}{\sqrt{a+b \sin[e+fx]}} dx \text{ when } a^2 - b^2 = 0$$

$$1. \int \frac{(g \cos[e+fx])^p}{\sqrt{a+b \sin[e+fx]}} dx \text{ when } a^2 - b^2 = 0 \wedge p > 0$$

$$1: \int \frac{\sqrt{g \cos[e+fx]}}{\sqrt{a+b \sin[e+fx]}} dx \text{ when } a^2 - b^2 = 0$$

Derivation: Piecewise constant extraction and algebraic expansion

Basis: If  $a^2 - b^2 = 0$ , then  $\partial_x \frac{\sqrt{1+\cos[e+fx]} \sqrt{a+b \sin[e+fx]}}{a+a \cos[e+fx]+b \sin[e+fx]} = 0$

Rule: If  $a^2 - b^2 = 0$ , then

$$\begin{aligned} & \int \frac{\sqrt{g \cos[e+fx]}}{\sqrt{a+b \sin[e+fx]}} dx \rightarrow \frac{g \sqrt{1+\cos[e+fx]} \sqrt{a+b \sin[e+fx]}}{a(a+a \cos[e+fx]+b \sin[e+fx])} \int \frac{a+a \cos[e+fx]-b \sin[e+fx]}{\sqrt{g \cos[e+fx]} \sqrt{1+\cos[e+fx]}} dx \\ & \rightarrow \frac{g \sqrt{1+\cos[e+fx]} \sqrt{a+b \sin[e+fx]}}{a+a \cos[e+fx]+b \sin[e+fx]} \int \frac{\sqrt{1+\cos[e+fx]}}{\sqrt{g \cos[e+fx]}} dx - \frac{g \sqrt{1+\cos[e+fx]} \sqrt{a+b \sin[e+fx]}}{b+b \cos[e+fx]+a \sin[e+fx]} \int \frac{\sin[e+fx]}{\sqrt{g \cos[e+fx]} \sqrt{1+\cos[e+fx]}} dx \end{aligned}$$

Program code:

```
Int[Sqrt[g_.cos[e_.+f_.x_]]/Sqrt[a+b_.sin[e_.+f_.x_]],x_Symbol] :=
  g*Sqrt[1+Cos[e+f*x]]*Sqrt[a+b*sin[e+f*x]]/(a+a*cos[e+f*x]+b*sin[e+f*x])*Int[Sqrt[1+Cos[e+f*x]]/Sqrt[g*cos[e+f*x]],x] -
  g*Sqrt[1+Cos[e+f*x]]*Sqrt[a+b*sin[e+f*x]]/(b+b*cos[e+f*x]+a*sin[e+f*x])*Int[Sin[e+f*x]/(Sqrt[g*cos[e+f*x]]*Sqrt[1+Cos[e+f*x]]),x]
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0]
```

**2:**  $\int \frac{(g \cos[e+fx])^{3/2}}{\sqrt{a+b \sin[e+fx]}} dx$  when  $a^2 - b^2 = 0$

Derivation: Symmetric cosine/sine recurrence 2a and 1c

Rule: If  $a^2 - b^2 = 0$ , then

$$\int \frac{(g \cos[e+fx])^{3/2}}{\sqrt{a+b \sin[e+fx]}} dx \rightarrow \frac{g \sqrt{g \cos[e+fx]} \sqrt{a+b \sin[e+fx]}}{b f} + \frac{g^2}{2 a} \int \frac{\sqrt{a+b \sin[e+fx]}}{\sqrt{g \cos[e+fx]}} dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^(3/2)/Sqrt[a+b_.sin[e_.+f_.x_]],x_Symbol] :=
  g*Sqrt[g*cos[e+f*x]]*Sqrt[a+b*sin[e+f*x]]/(b*f) +
  g^2/(2*a)*Int[Sqrt[a+b*sin[e+f*x]]/Sqrt[g*cos[e+f*x]],x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0]
```



**3:**  $\int \frac{(g \cos[e+fx])^p}{\sqrt{a+b \sin[e+fx]}} dx$  when  $a^2 - b^2 = 0 \wedge p > 2$

Derivation: Symmetric cosine/sine recurrence 1c with  $n \rightarrow -\frac{1}{2}$

Rule: If  $a^2 - b^2 = 0 \wedge p > 2$ , then

$$\int \frac{(g \cos[e+fx])^p}{\sqrt{a+b \sin[e+fx]}} dx \rightarrow -\frac{2b (g \cos[e+fx])^{p+1}}{fg(2p-1)(a+b \sin[e+fx])^{3/2}} + \frac{2a(p-2)}{2p-1} \int \frac{(g \cos[e+fx])^p}{(a+b \sin[e+fx])^{3/2}} dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_/Sqrt[a_+b_.sin[e_.+f_.x_]],x_Symbol] :=
  -2*b*(g*cos[e+f*x])^(p+1)/(f*g*(2*p-1)*(a+b*sin[e+f*x])^(3/2)) +
  2*a*(p-2)/(2*p-1)*Int[(g*cos[e+f*x])^p/(a+b*sin[e+f*x])^(3/2),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && GtQ[p,2] && IntegerQ[2*p]
```

**2:**  $\int \frac{(g \cos[e+fx])^p}{\sqrt{a+b \sin[e+fx]}} dx$  when  $a^2 - b^2 = 0 \wedge p < -1$

Derivation: Symmetric cosine/sine recurrence 1b with  $n \rightarrow -\frac{1}{2}$

Rule: If  $a^2 - b^2 = 0 \wedge p < -1$ , then

$$\int \frac{(g \cos[e+fx])^p}{\sqrt{a+b \sin[e+fx]}} dx \rightarrow -\frac{b (g \cos[e+fx])^{p+1}}{a f g (p+1) \sqrt{a+b \sin[e+fx]}} + \frac{a (2p+1)}{2 g^2 (p+1)} \int \frac{(g \cos[e+fx])^{p+2}}{(a+b \sin[e+fx])^{3/2}} dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_/Sqrt[a_+b_.sin[e_.+f_.x_]],x_Symbol] :=
  -b*(g*cos[e+f*x])^(p+1)/(a*f*g*(p+1)*Sqrt[a+b*sin[e+f*x]]) +
  a*(2*p+1)/(2*g^2*(p+1))*Int[(g*cos[e+f*x])^(p+2)/(a+b*sin[e+f*x])^(3/2),x] /;
FreeQ[{a,b,e,f,g},x] && EqQ[a^2-b^2,0] && LtQ[p,-1] && IntegerQ[2*p]
```

$$8. \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0$$

$$1: \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge m \in \mathbb{Z}$$

Derivation: Piecewise constant extraction and integration by substitution

$$\text{Basis: If } a^2 - b^2 = 0, \text{ then } \partial_x \frac{(g \cos[e+fx])^{p+1}}{(1+\sin[e+fx])^{\frac{p+1}{2}} (1-\sin[e+fx])^{\frac{p+1}{2}}} = 0$$

$$\text{Basis: If } a^2 - b^2 = 0, \text{ then } \frac{(g \cos[e+fx])^{p+1}}{g (1+\sin[e+fx])^{\frac{p+1}{2}} (1-\sin[e+fx])^{\frac{p+1}{2}}} \frac{\cos[e+fx] (1+\frac{b}{a} \sin[e+fx])^{\frac{p-1}{2}} (1-\frac{b}{a} \sin[e+fx])^{\frac{p-1}{2}}}{(g \cos[e+fx])^p} = 1$$

$$\text{Basis: } \cos[e+fx] = \frac{1}{f} \partial_x \sin[e+fx]$$

Rule: If  $a^2 - b^2 = 0 \wedge m \in \mathbb{Z}$ , then

$$\begin{aligned} & \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow a^m \int (g \cos[e+fx])^p \left(1 + \frac{b}{a} \sin[e+fx]\right)^m dx \rightarrow \\ & \frac{a^m (g \cos[e+fx])^{p+1}}{g (1+\sin[e+fx])^{\frac{p+1}{2}} (1-\sin[e+fx])^{\frac{p+1}{2}}} \int \cos[e+fx] \left(1 + \frac{b}{a} \sin[e+fx]\right)^{m+\frac{p-1}{2}} \left(1 - \frac{b}{a} \sin[e+fx]\right)^{\frac{p-1}{2}} dx \rightarrow \\ & \frac{a^m (g \cos[e+fx])^{p+1}}{f g (1+\sin[e+fx])^{\frac{p+1}{2}} (1-\sin[e+fx])^{\frac{p+1}{2}}} \text{Subst}\left[\int \left(1 + \frac{b}{a} x\right)^{m+\frac{p-1}{2}} \left(1 - \frac{b}{a} x\right)^{\frac{p-1}{2}} dx, x, \sin[e+fx]\right] \end{aligned}$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  a^m*(g*cos[e+f*x])^(p+1)/(f*g*(1+Sin[e+f*x])^((p+1)/2)*(1-Sin[e+f*x])^((p+1)/2))*
  Subst[Int[(1+b/a*x)^(m+(p-1)/2)*(1-b/a*x)^((p-1)/2),x],x,Sin[e+f*x]] /;
FreeQ[{a,b,e,f,g,p},x] && EqQ[a^2-b^2,0] && IntegerQ[m]
```

$$\mathbf{2:} \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 = 0 \wedge m \notin \mathbb{Z}$$

Derivation: Piecewise constant extraction and integration by substitution

$$\text{Basis: If } a^2 - b^2 = 0, \text{ then } \partial_x \frac{(g \cos[e+fx])^{p+1}}{(a+b \sin[e+fx])^{\frac{p+1}{2}} (a-b \sin[e+fx])^{\frac{p+1}{2}}} = 0$$

$$\text{Basis: If } a^2 - b^2 = 0, \text{ then } \frac{a^2 (g \cos[e+fx])^{p+1}}{g (a+b \sin[e+fx])^{\frac{p+1}{2}} (a-b \sin[e+fx])^{\frac{p+1}{2}}} \frac{\cos[e+fx] (a+b \sin[e+fx])^{\frac{p-1}{2}} (a-b \sin[e+fx])^{\frac{p-1}{2}}}{(g \cos[e+fx])^p} = 1$$

$$\text{Basis: } \cos[e+fx] = \frac{1}{f} \partial_x \sin[e+fx]$$

Rule: If  $a^2 - b^2 = 0 \wedge m \notin \mathbb{Z}$ , then

$$\begin{aligned} & \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \\ & \frac{a^2 (g \cos[e+fx])^{p+1}}{g (a+b \sin[e+fx])^{\frac{p+1}{2}} (a-b \sin[e+fx])^{\frac{p+1}{2}}} \int \cos[e+fx] (a+b \sin[e+fx])^{m+\frac{p-1}{2}} (a-b \sin[e+fx])^{\frac{p-1}{2}} dx \rightarrow \\ & \frac{a^2 (g \cos[e+fx])^{p+1}}{f g (a+b \sin[e+fx])^{\frac{p+1}{2}} (a-b \sin[e+fx])^{\frac{p+1}{2}}} \text{Subst}\left[\int (a+b x)^{m+\frac{p-1}{2}} (a-b x)^{\frac{p-1}{2}} dx, x, \sin[e+fx]\right] \end{aligned}$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a+_b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  a^2*(g*cos[e+f*x])^(p+1)/(f*g*(a+b*sin[e+f*x])^((p+1)/2)*(a-b*sin[e+f*x])^((p+1)/2))*
  Subst[Int[(a+b*x)^(m+(p-1)/2)*(a-b*x)^((p-1)/2),x],x,Sin[e+f*x]] /;
FreeQ[{a,b,e,f,g,m,p},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[m]]
```

4.  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 \neq 0$

1.  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 \neq 0 \wedge m > 0$

1.  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 \neq 0 \wedge m > 0 \wedge p < -1$

**1:**  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 \neq 0 \wedge 0 < m < 1 \wedge p < -1$

Derivation: Nondegenerate sine recurrence 3a with  $c \rightarrow 1$ ,  $d \rightarrow 0$ ,  $A \rightarrow 1$ ,  $B \rightarrow 0$ ,  $C \rightarrow 0$

Derivation: Nondegenerate sine recurrence 3b with  $c \rightarrow 0$ ,  $d \rightarrow 1$ ,  $A \rightarrow 0$ ,  $B \rightarrow a$ ,  $C \rightarrow b$ ,  $m \rightarrow m-1$ ,  $n \rightarrow -1$

Derivation: Nondegenerate sine recurrence 3a with  $c \rightarrow 0$ ,  $d \rightarrow 1$ ,  $A \rightarrow 0$ ,  $B \rightarrow 1$ ,  $C \rightarrow 0$ ,  $n \rightarrow -1$

Rule: If  $a^2 - b^2 \neq 0 \wedge 0 < m < 1 \wedge p < -1$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow$$

$$- \frac{(g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^m \sin[e+fx]}{f g (p+1)} +$$

$$\frac{1}{g^2 (p+1)} \int (g \cos[e+fx])^{p+2} (a+b \sin[e+fx])^{m-1} (a(p+2) + b(m+p+2) \sin[e+fx]) dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  -(g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^m*sin[e+f*x]/(f*g*(p+1)) +
  1/(g^2*(p+1))*Int[(g*cos[e+f*x])^(p+2)*(a+b*sin[e+f*x])^(m-1)*(a*(p+2)+b*(m+p+2)*sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0] && LtQ[0,m,1] && LtQ[p,-1] && (IntegersQ[2*m,2*p] || IntegerQ[m])
```

**2:**  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 \neq 0 \wedge m > 1 \wedge p < -1$

Derivation: Nondegenerate sine recurrence 3a with  $c \rightarrow 0$ ,  $d \rightarrow 1$ ,  $A \rightarrow 0$ ,  $B \rightarrow a$ ,  $C \rightarrow b$ ,  $m \rightarrow m-1$ ,  $n \rightarrow -1$

Rule: If  $a^2 - b^2 \neq 0 \wedge m > 1 \wedge p < -1$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow$$

$$- \frac{(g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^{m-1} (b+a \sin[e+fx])}{f g (p+1)} +$$

$$\frac{1}{g^2 (p+1)} \int (g \cos[e+fx])^{p+2} (a+b \sin[e+fx])^{m-2} (b^2 (m-1) + a^2 (p+2) + a b (m+p+1) \sin[e+fx]) dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a_+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  -(g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^(m-1)*(b+a*sin[e+f*x])/(f*g*(p+1)) +
  1/(g^2*(p+1))*Int[(g*cos[e+f*x])^(p+2)*(a+b*sin[e+f*x])^(m-2)*(b^2*(m-1)+a^2*(p+2)+a*b*(m+p+1)*sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0] && GtQ[m,1] && LtQ[p,-1] && (IntegersQ[2*m,2*p] || IntegerQ[m])
```

**2:**  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 \neq 0 \wedge m > 1 \wedge m+p \neq 0$

Derivation: Nondegenerate sine recurrence 1b with  $c \rightarrow 0$ ,  $d \rightarrow 1$ ,  $A \rightarrow 0$ ,  $B \rightarrow a$ ,  $C \rightarrow b$ ,  $m \rightarrow m-1$ ,  $n \rightarrow -1$

Rule: If  $a^2 - b^2 \neq 0 \wedge m > 1 \wedge m+p \neq 0$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow$$

$$- \frac{b (g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^{m-1}}{f g (m+p)} +$$

$$\frac{1}{m+p} \int (g \cos[e+fx])^p (a+b \sin[e+fx])^{m-2} (b^2 (m-1) + a^2 (m+p) + a b (2m+p-1) \sin[e+fx]) dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a_+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  -b*(g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^(m-1)/(f*g*(m+p)) +
  1/(m+p)*Int[(g*cos[e+f*x])^p*(a+b*sin[e+f*x])^(m-2)*(b^2*(m-1)+a^2*(m+p)+a*b*(2*m+p-1)*Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,p},x] && NeQ[a^2-b^2,0] && GtQ[m,1] && NeQ[m+p,0] && (IntegersQ[2*m,2*p] || IntegerQ[m])
```

$$2. \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 \neq 0 \wedge m < -1$$

$$1: \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 \neq 0 \wedge m < -1 \wedge p > 1$$

Derivation: Nondegenerate sine recurrence 2a with  $c \rightarrow 0$ ,  $d \rightarrow 1$ ,  $A \rightarrow 0$ ,  $B \rightarrow 1$ ,  $C \rightarrow 0$ ,  $n \rightarrow -1$

Derivation: Integration by parts

■

$$\text{Basis: } \cos[e+fx] (a+b \sin[e+fx])^n = \partial_x \frac{(a+b \sin[e+fx])^{n+1}}{b f (n+1)}$$

Rule: If  $a^2 - b^2 \neq 0 \wedge m < -1 \wedge p > 1$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \frac{g (g \cos[e+fx])^{p-1} (a+b \sin[e+fx])^{m+1}}{b f (m+1)} + \frac{g^2 (p-1)}{b (m+1)} \int (g \cos[e+fx])^{p-2} (a+b \sin[e+fx])^{m+1} \sin[e+fx] dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  g*(g*cos[e+f*x])^(p-1)*(a+b*sin[e+f*x])^(m+1)/(b*f*(m+1)) +
  g^2*(p-1)/(b*(m+1))*Int[(g*cos[e+f*x])^(p-2)*(a+b*sin[e+f*x])^(m+1)*sin[e+f*x],x] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && GtQ[p,1] && IntegersQ[2*m,2*p]
```



$$\mathbf{2:} \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 \neq 0 \wedge m < -1$$

Derivation: Nondegenerate sine recurrence 1a with  $c \rightarrow 1$ ,  $d \rightarrow 0$ ,  $A \rightarrow 1$ ,  $B \rightarrow 0$ ,  $C \rightarrow 0$

Derivation: Nondegenerate sine recurrence 1c with  $c \rightarrow 1$ ,  $d \rightarrow 0$ ,  $A \rightarrow 1$ ,  $B \rightarrow 0$ ,  $C \rightarrow 0$

Derivation: Nondegenerate sine recurrence 1c with  $c \rightarrow 0$ ,  $d \rightarrow 1$ ,  $A \rightarrow 0$ ,  $B \rightarrow 1$ ,  $C \rightarrow 0$ ,  $n \rightarrow -1$

Rule: If  $a^2 - b^2 \neq 0 \wedge m < -1$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow -\frac{b (g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^{m+1}}{f g (a^2 - b^2) (m+1)} + \frac{1}{(a^2 - b^2) (m+1)} \int (g \cos[e+fx])^p (a+b \sin[e+fx])^{m+1} (a(m+1) - b(m+p+2) \sin[e+fx]) dx$$

Program code:

```
Int[(g_.*cos[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  -b*(g*cos[e+fx])^(p+1)*(a+b*sin[e+fx])^(m+1)/(f*g*(a^2-b^2)*(m+1)) +
  1/((a^2-b^2)*(m+1))*Int[(g*cos[e+fx])^p*(a+b*sin[e+fx])^(m+1)*(a*(m+1)-b*(m+p+2)*sin[e+fx]),x] /;
FreeQ[{a,b,e,f,g,p},x] && NeQ[a^2-b^2,0] && LtQ[m,-1] && IntegersQ[2*m,2*p]
```

$$\mathbf{3:} \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 \neq 0 \wedge p > 1 \wedge m+p \neq 0$$

Derivation: Nondegenerate sine recurrence 2a with  $c \rightarrow 0$ ,  $d \rightarrow 1$ ,  $A \rightarrow 0$ ,  $B \rightarrow a$ ,  $C \rightarrow b$ ,  $m \rightarrow m-1$ ,  $n \rightarrow -1$

Derivation: Nondegenerate sine recurrence 2b with  $c \rightarrow 0$ ,  $d \rightarrow 1$ ,  $A \rightarrow 0$ ,  $B \rightarrow 1$ ,  $C \rightarrow 0$ ,  $n \rightarrow -1$

Rule: If  $a^2 - b^2 \neq 0 \wedge p > 1 \wedge m+p \neq 0$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow$$

$$\frac{g (g \cos[e+fx])^{p-1} (a+b \sin[e+fx])^{m+1}}{b f (m+p)} + \frac{g^2 (p-1)}{b (m+p)} \int (g \cos[e+fx])^{p-2} (a+b \sin[e+fx])^m (b+a \sin[e+fx]) dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a_+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  g*(g*cos[e+f*x])^(p-1)*(a+b*sin[e+f*x])^(m+1)/(b*f*(m+p)) +
  g^2*(p-1)/(b*(m+p))*Int[(g*cos[e+f*x])^(p-2)*(a+b*sin[e+f*x])^m*(b+a*sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,m},x] && NeQ[a^2-b^2,0] && GtQ[p,1] && NeQ[m+p,0] && IntegersQ[2*m,2*p]
```

4:  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 \neq 0 \wedge p < -1$

Derivation: Nondegenerate sine recurrence 3b with  $c \rightarrow 1$ ,  $d \rightarrow 0$ ,  $A \rightarrow 1$ ,  $B \rightarrow 0$ ,  $C \rightarrow 0$

Derivation: Nondegenerate sine recurrence 3b with  $c \rightarrow 0$ ,  $d \rightarrow 1$ ,  $A \rightarrow 0$ ,  $B \rightarrow 1$ ,  $C \rightarrow 0$ ,  $n \rightarrow -1$

Rule: If  $a^2 - b^2 \neq 0 \wedge p < -1$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow$$

$$\frac{(g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^{m+1} (b-a \sin[e+fx])}{f g (a^2 - b^2) (p+1)} +$$

$$\frac{1}{g^2 (a^2 - b^2) (p+1)} \int (g \cos[e+fx])^{p+2} (a+b \sin[e+fx])^m (a^2 (p+2) - b^2 (m+p+2) + a b (m+p+3) \sin[e+fx]) dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a_+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  (g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^(m+1)*(b-a*sin[e+f*x])/(f*g*(a^2-b^2)*(p+1)) +
  1/(g^2*(a^2-b^2)*(p+1))*Int[(g*cos[e+f*x])^(p+2)*(a+b*sin[e+f*x])^m*(a^2*(p+2)-b^2*(m+p+2)+a*b*(m+p+3)*sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,m},x] && NeQ[a^2-b^2,0] && LtQ[p,-1] && IntegersQ[2*m,2*p]
```

$$5. \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 \neq 0 \wedge m+p \in \mathbb{Z}^-$$

$$1. \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 \neq 0 \wedge m+p+1 = 0$$

$$1: \int \frac{1}{\sqrt{g \cos[e+fx]} \sqrt{a+b \sin[e+fx]}} dx \text{ when } a^2 - b^2 \neq 0$$

Derivation: Piecewise constant extraction and integration by substitution

$$\text{Basis: } \partial_x \frac{\sqrt{g \cos[e+fx]} \sqrt{\frac{a+b \sin[e+fx]}{(a-b)(1-\sin[e+fx])}}}{\sqrt{a+b \sin[e+fx]} \sqrt{\frac{1+\cos[e+fx]+\sin[e+fx]}{1+\cos[e+fx]-\sin[e+fx]}}} = 0$$

$$\text{Basis: } \frac{\sqrt{\frac{a+b \sin[e+fx]}{(a-b)(1-\sin[e+fx])}}}{(a+b \sin[e+fx]) \sqrt{\frac{1+\cos[e+fx]+\sin[e+fx]}{1+\cos[e+fx]-\sin[e+fx]}}} =$$

$$\frac{2\sqrt{2}}{(a-b)f} \text{Subst} \left[ \frac{1}{\sqrt{1+\frac{(a+b)x^4}{a-b}}}, x, \sqrt{\frac{1+\cos[e+fx]+\sin[e+fx]}{1+\cos[e+fx]-\sin[e+fx]}} \right] \partial_x \sqrt{\frac{1+\cos[e+fx]+\sin[e+fx]}{1+\cos[e+fx]-\sin[e+fx]}}$$

Rule: If  $a^2 - b^2 \neq 0$ , then

$$\int \frac{1}{\sqrt{g \cos[e+fx]} \sqrt{a+b \sin[e+fx]}} dx \rightarrow$$

$$\frac{(a-b) \sqrt{g \cos[e+fx]} \sqrt{\frac{a+b \sin[e+fx]}{(a-b)(1-\sin[e+fx])}}}{g \sqrt{a+b \sin[e+fx]} \sqrt{\frac{1+\cos[e+fx]+\sin[e+fx]}{1+\cos[e+fx]-\sin[e+fx]}}} \int \frac{\sqrt{\frac{a+b \sin[e+fx]}{(a-b)(1-\sin[e+fx])}}}{(a+b \sin[e+fx]) \sqrt{\frac{1+\cos[e+fx]+\sin[e+fx]}{1+\cos[e+fx]-\sin[e+fx]}}} dx \rightarrow$$

$$\frac{2 \sqrt{2} \sqrt{g \cos[e+fx]} \sqrt{\frac{a+b \sin[e+fx]}{(a-b)(1-\sin[e+fx])}}}{f g \sqrt{a+b \sin[e+fx]} \sqrt{\frac{1+\cos[e+fx]+\sin[e+fx]}{1+\cos[e+fx]-\sin[e+fx]}}} \text{Subst}\left[\int \frac{1}{\sqrt{1+\frac{(a+b)x^4}{a-b}}} dx, x, \sqrt{\frac{1+\cos[e+fx]+\sin[e+fx]}{1+\cos[e+fx]-\sin[e+fx]}}\right]$$

Program code:

```
Int[1/(Sqrt[g_.*cos[e_.+f_.*x_]]*Sqrt[a_+b_.*sin[e_.+f_.*x_]]),x_Symbol] :=
  2*Sqrt[2]*Sqrt[g*Cos[e+f*x]]*Sqrt[(a+b*Sin[e+f*x])/((a-b)*(1-Sin[e+f*x]))]/
  (f*g*Sqrt[a+b*Sin[e+f*x]]*Sqrt[(1+Cos[e+f*x]+Sin[e+f*x])/(1+Cos[e+f*x]-Sin[e+f*x])]) *
  Subst[Int[1/Sqrt[1+(a+b)*x^4/(a-b)],x],x,Sqrt[(1+Cos[e+f*x]+Sin[e+f*x])/(1+Cos[e+f*x]-Sin[e+f*x])]] /;
FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0]
```

**2:**  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 \neq 0 \wedge m+p+1 = 0$

Derivation: Integration by substitution

Rule: If  $a^2 - b^2 \neq 0 \wedge m+p+1 = 0$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow$$

$$\frac{1}{f(a+b)^{m+1}} g (g \cos[e+fx])^{p-1} (1 - \sin[e+fx]) (a+b \sin[e+fx])^{m+1} \left( -\frac{(a-b)(1 - \sin[e+fx])}{(a+b)(1 + \sin[e+fx])} \right)^{\frac{m}{2}}$$

$$\text{Hypergeometric2F1}\left[m+1, \frac{m}{2}+1, m+2, \frac{2(a+b \sin[e+fx])}{(a+b)(1 + \sin[e+fx])}\right]$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  g*(g*cos[e+f*x])^(p-1)*(1-Sin[e+f*x])*(a+b*sin[e+f*x])^(m+1)*(-(a-b)*(1-Sin[e+f*x])/((a+b)*(1+Sin[e+f*x])))^(m/2)/
  (f*(a+b)^(m+1))*
  Hypergeometric2F1[m+1,m/2+1,m+2,2*(a+b*sin[e+f*x])/((a+b)*(1+Sin[e+f*x]))] /;
FreeQ[{a,b,e,f,g,m,p},x] && NeQ[a^2-b^2,0] && EqQ[m+p+1,0]
```

**2:**  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 \neq 0 \wedge m+p+2 \neq 0$

Rule: If  $a^2 - b^2 \neq 0 \wedge m+p+2 \neq 0$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \frac{(g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^{m+1}}{f g (a-b) (p+1)} + \frac{a}{g^2 (a-b)} \int \frac{(g \cos[e+fx])^{p+2} (a+b \sin[e+fx])^m}{1 - \sin[e+fx]} dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  (g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^(m+1)/(f*g*(a-b)*(p+1)) +
  a/(g^2*(a-b))*Int[(g*cos[e+f*x])^(p+2)*(a+b*sin[e+f*x])^m/(1-Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && NeQ[a^2-b^2,0] && EqQ[m+p+2,0]
```

**3:**  $\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $a^2 - b^2 \neq 0 \wedge m+p+2 \in \mathbb{Z}^-$

Rule: If  $a^2 - b^2 \neq 0 \wedge m+p+2 \in \mathbb{Z}^-$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \frac{(g \cos[e+fx])^{p+1} (a+b \sin[e+fx])^{m+1}}{f g (a-b) (p+1)} - \frac{b (m+p+2)}{g^2 (a-b) (p+1)} \int (g \cos[e+fx])^{p+2} (a+b \sin[e+fx])^m dx + \frac{a}{g^2 (a-b)} \int \frac{(g \cos[e+fx])^{p+2} (a+b \sin[e+fx])^m}{1 - \sin[e+fx]} dx$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  (g*cos[e+f*x])^(p+1)*(a+b*sin[e+f*x])^(m+1)/(f*g*(a-b)*(p+1)) -
  b*(m+p+2)/(g^2*(a-b)*(p+1))*Int[(g*cos[e+f*x])^(p+2)*(a+b*sin[e+f*x])^m,x] +
  a/(g^2*(a-b))*Int[(g*cos[e+f*x])^(p+2)*(a+b*sin[e+f*x])^m/(1-Sin[e+f*x]),x] /;
FreeQ[{a,b,e,f,g,m,p},x] && NeQ[a^2-b^2,0] && ILtQ[m+p+2,0]
```

6:  $\int \frac{\sqrt{g \cos[e+fx]}}{a+b \sin[e+fx]} dx$  when  $a^2 - b^2 \neq 0$

Derivation: Algebraic expansion and integration by substitution

Basis:  $\frac{1}{a+b \sin[z]} = \frac{a-b \sin[z]}{a^2-b^2 \sin^2[z]} = \frac{a}{a^2-b^2+b^2 \cos^2[z]} - \frac{b \sin[z]}{a^2-b^2+b^2 \cos^2[z]}$

Basis: Let  $q = \sqrt{-a^2 + b^2}$ , then  $\frac{\sqrt{g \cos[z]}}{a^2-b^2+b^2 \cos^2[z]} = \frac{g}{2b \sqrt{g \cos[z]} (q+b \cos[z])} - \frac{g}{2b \sqrt{g \cos[z]} (q-b \cos[z])}$

Basis:  $\sin[e+fx] F[g \cos[e+fx]] = -\frac{1}{fg} \text{Subst}[F[x], x, g \cos[e+fx]] \partial_x (g \cos[e+fx])$

Rule: If  $a^2 - b^2 \neq 0$ , let  $q = \sqrt{-a^2 + b^2}$ , then

$$\int \frac{\sqrt{g \cos[e+fx]}}{a+b \sin[e+fx]} dx \rightarrow a \int \frac{\sqrt{g \cos[e+fx]}}{a^2-b^2+b^2 \cos^2[e+fx]} dx - b \int \frac{\sin[e+fx] \sqrt{g \cos[e+fx]}}{a^2-b^2+b^2 \cos^2[e+fx]} dx$$

$$\rightarrow \frac{a g}{2 b} \int \frac{1}{\sqrt{g \cos[e+fx]} (q+b \cos[e+fx])} dx - \frac{a g}{2 b} \int \frac{1}{\sqrt{g \cos[e+fx]} (q-b \cos[e+fx])} dx + \frac{b g}{f} \text{Subst}\left[\int \frac{\sqrt{x}}{g^2 (a^2-b^2) + b^2 x^2} dx, x, g \cos[e+fx]\right]$$

Program code:

```
Int[Sqrt[g_.cos[e_.+f_.*x_]]/(a_.+b_.sin[e_.+f_.*x_]),x_Symbol] :=
  With[{q=Rt[-a^2+b^2,2]},
    a*g/(2*b)*Int[1/(Sqrt[g*cos[e+f*x]]*(q+b*cos[e+f*x])),x] -
    a*g/(2*b)*Int[1/(Sqrt[g*cos[e+f*x]]*(q-b*cos[e+f*x])),x] +
    b*g/f*Subst[Int[Sqrt[x]/(g^2*(a^2-b^2)+b^2*x^2),x],x,g*cos[e+f*x]] /;
  FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0]
```



7:  $\int \frac{1}{\sqrt{g \cos[e+fx]} (a+b \sin[e+fx])} dx$  when  $a^2 - b^2 \neq 0$

Derivation: Algebraic expansion and integration by substitution

Basis:  $\frac{1}{a+b \sin[z]} = \frac{a-b \sin[z]}{a^2-b^2 \sin^2[z]} = \frac{a}{a^2-b^2+b^2 \cos^2[z]} - \frac{b \sin[z]}{a^2-b^2+b^2 \cos^2[z]}$

Basis: Let  $q = \sqrt{-a^2 + b^2}$ , then  $\frac{1}{a^2-b^2+b^2 \cos^2[z]} = -\frac{1}{2q(q+b \cos[z])} - \frac{1}{2q(q-b \cos[z])}$

Basis:  $\sin[e+fx] F[g \cos[e+fx]] = -\frac{1}{fg} \text{Subst}[F[x], x, g \cos[e+fx]] \partial_x (g \cos[e+fx])$

■ Rule: If  $a^2 - b^2 \neq 0$ , let  $q = \sqrt{-a^2 + b^2}$ , then

$$\begin{aligned} \int \frac{1}{\sqrt{g \cos[e+fx]} (a+b \sin[e+fx])} dx &\rightarrow a \int \frac{1}{\sqrt{g \cos[e+fx]} (a^2 - b^2 + b^2 \cos[e+fx]^2)} dx - b \int \frac{\sin[e+fx]}{\sqrt{g \cos[e+fx]} (a^2 - b^2 + b^2 \cos[e+fx]^2)} dx \\ &\rightarrow -\frac{a}{2q} \int \frac{1}{\sqrt{g \cos[e+fx]} (q+b \cos[e+fx])} dx - \\ &\quad \frac{a}{2q} \int \frac{1}{\sqrt{g \cos[e+fx]} (q-b \cos[e+fx])} dx + \frac{bg}{f} \text{Subst}\left[\int \frac{1}{\sqrt{x} (g^2 (a^2 - b^2) + b^2 x^2)} dx, x, g \cos[e+fx]\right] \end{aligned}$$

Program code:

```
Int[1/(Sqrt[g_.cos[e_.+f_.x_]]*(a_+b_.sin[e_.+f_.x_])),x_Symbol] :=
  With[{q=Rt[-a^2+b^2,2]},
    -a/(2*q)*Int[1/(Sqrt[g*cos[e+f*x]]*(q+b*cos[e+f*x])),x] -
    a/(2*q)*Int[1/(Sqrt[g*cos[e+f*x]]*(q-b*cos[e+f*x])),x] +
    b*g/f*Subst[Int[1/(Sqrt[x]*(g^2*(a^2-b^2)+b^2*x^2)),x],x,g*cos[e+f*x]] /;
  FreeQ[{a,b,e,f,g},x] && NeQ[a^2-b^2,0]
```

$$8. \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 \neq 0 \wedge m \notin \mathbb{Z}^+$$

$$1: \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 \neq 0 \wedge m \in \mathbb{Z}^- \wedge m+p+1 \notin \mathbb{Z}^+$$

Derivation: Integration by substitution

Rule: If  $a^2 - b^2 \neq 0 \wedge m \in \mathbb{Z}^- \wedge m+p+1 \notin \mathbb{Z}^+$ , then

$$\int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \frac{g (g \cos[e+fx])^{p-1} (a+b \sin[e+fx])^{m+1}}{b f (m+p) \left( -\frac{b(1-\sin[e+fx])}{a+b \sin[e+fx]} \right)^{\frac{p-1}{2}} \left( \frac{b(1+\sin[e+fx])}{a+b \sin[e+fx]} \right)^{\frac{p-1}{2}}} \text{AppellF1} \left[ -p-m, \frac{1-p}{2}, \frac{1-p}{2}, 1-p-m, \frac{a+b}{a+b \sin[e+fx]}, \frac{a-b}{a+b \sin[e+fx]} \right]$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  g*(g*cos[e+f*x])^(p-1)*(a+b*sin[e+f*x])^(m+1)/
  (b*f*(m+p)*(-b*(1-Sin[e+f*x])/(a+b*sin[e+f*x]))^((p-1)/2)*(b*(1+Sin[e+f*x])/(a+b*sin[e+f*x]))^((p-1)/2))*
  AppellF1[-p-m,(1-p)/2,(1-p)/2,1-p-m,(a+b)/(a+b*sin[e+f*x]),(a-b)/(a+b*sin[e+f*x])] /;
FreeQ[{a,b,e,f,g,p},x] && NeQ[a^2-b^2,0] && ILtQ[m,0] && Not[IGtQ[m+p+1,0]]
```

$$\mathbf{2:} \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \text{ when } a^2 - b^2 \neq 0 \wedge m \notin \mathbb{Z}^+$$

Derivation: Piecewise constant extraction and integration by substitution

$$\text{Basis: } \partial_x \frac{(g \cos[e+fx])^{p-1}}{\left(1 - \frac{a+b \sin[e+fx]}{a-b}\right)^{\frac{p-1}{2}} \left(1 - \frac{a+b \sin[e+fx]}{a+b}\right)^{\frac{p-1}{2}}} = 0$$

$$\text{Basis: } \cos[e+fx] = \frac{1}{f} \partial_x \sin[e+fx]$$

Rule: If  $a^2 - b^2 \neq 0 \wedge m \notin \mathbb{Z}^+$ , then

$$\begin{aligned} & \int (g \cos[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow \\ & \frac{g (g \cos[e+fx])^{p-1}}{\left(1 - \frac{a+b \sin[e+fx]}{a-b}\right)^{\frac{p-1}{2}} \left(1 - \frac{a+b \sin[e+fx]}{a+b}\right)^{\frac{p-1}{2}}} \int \cos[e+fx] \left(-\frac{b}{a-b} - \frac{b \sin[e+fx]}{a-b}\right)^{\frac{p-1}{2}} \left(\frac{b}{a+b} - \frac{b \sin[e+fx]}{a+b}\right)^{\frac{p-1}{2}} (a+b \sin[e+fx])^m dx \rightarrow \\ & \frac{g (g \cos[e+fx])^{p-1}}{f \left(1 - \frac{a+b \sin[e+fx]}{a-b}\right)^{\frac{p-1}{2}} \left(1 - \frac{a+b \sin[e+fx]}{a+b}\right)^{\frac{p-1}{2}}} \text{Subst} \left[ \int \left(-\frac{b}{a-b} - \frac{bx}{a-b}\right)^{\frac{p-1}{2}} \left(\frac{b}{a+b} - \frac{bx}{a+b}\right)^{\frac{p-1}{2}} (a+bx)^m dx, x, \sin[e+fx] \right] \end{aligned}$$

Program code:

```
Int[(g_.cos[e_.+f_.x_])^p_*(a+b_.sin[e_.+f_.x_])^m_,x_Symbol] :=
  g*(g*cos[e+f*x])^(p-1)/(f*(1-(a+b*sin[e+f*x])/(a-b))^(p-1)/2*(1-(a+b*sin[e+f*x])/(a+b))^(p-1)/2)*
  Subst[Int[(-b/(a-b)-b*x/(a-b))^(p-1)/2*(b/(a+b)-b*x/(a+b))^(p-1)/2*(a+b*x)^m,x],x,Sin[e+f*x]] /;
FreeQ[{a,b,e,f,g,m,p},x] && NeQ[a^2-b^2,0] && Not[IGtQ[m,0]]
```

### Rules for integrands of the form $(g \sec[e+fx])^p (a+b \sin[e+fx])^m$

**1:**  $\int (g \sec[e+fx])^p (a+b \sin[e+fx])^m dx$  when  $p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis:  $\partial_x ((g \cos[e+fx])^p (g \sec[e+fx])^p) = 0$

Rule: If  $p \notin \mathbb{Z}$ , then

$$\int (g \sec[e+fx])^p (a+b \sin[e+fx])^m dx \rightarrow g^{2 \text{IntPart}[p]} (g \cos[e+fx])^{\text{FracPart}[p]} (g \sec[e+fx])^{\text{FracPart}[p]} \int \frac{(a+b \sin[e+fx])^m}{(g \cos[e+fx])^p} dx$$

Program code:

```
Int[(g_.*sec[e_.+f_.*x_])^p_*(a_+b_.*sin[e_.+f_.*x_])^m_,x_Symbol] :=
  g^(2*IntPart[p])*(g*cos[e+f*x])^FracPart[p]*(g*sec[e+f*x])^FracPart[p]*Int[(a+b*sin[e+f*x])^m/(g*cos[e+f*x])^p,x] /;
FreeQ[{a,b,e,f,g,m,p},x] && Not[IntegerQ[p]]
```