Rules for integrands of the form $(c\,x)^m\,\left(a\,x^j + b\,x^n\right)^p$

1: $\int x^m\,\left(a\,x^j + b\,x^n\right)^p\,dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land m - n + 1 == 0$

Derivation: Integration by substitution

Basis: $x^{n-1}\,F[x^n] == \frac{1}{n}\,\text{Subst}[F[x],\,x,\,x^n]\,\partial_x x^n$

Rule: If $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land m - n + 1 == 0$, then

$$\int x^m\,\left(a\,x^j + b\,x^n\right)^p\,dx \;\longrightarrow\; \int x^m\,\left(a\,\left(x^n\right)^{j/n} + b\,x^n\right)^p\,dx \;\longrightarrow\; \frac{1}{n}\,\text{Subst}\left[\int\left(a\,x^{j/n} + b\,x\right)^p\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  1/n*Subst[Int[(a*x^Simplify[j/n]+b*x)^p,x],x,x^n] /;
FreeQ[{a,b,j,m,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && IntegerQ[Simplify[j/n]] && EqQ[Simplify[m-n+1],0]
```

2: $\int (c\,x)^m\,\left(a\,x^j + b\,x^n\right)^p\,dx$ when $p \notin \mathbb{Z} \land j \neq n \land m + n\,p + n - j + 1 == 0 \land (j \in \mathbb{Z} \lor c > 0)$

Derivation: Generalized binomial recurrence 2a

Rule: If $p \notin \mathbb{Z} \land j \neq n \land m + n\,p + n - j + 1 == 0 \land (j \in \mathbb{Z} \lor c > 0)$, then

$$\int (c\,x)^m\,\left(a\,x^j + b\,x^n\right)^p\,dx \;\longrightarrow\; -\frac{c^{j-1}\,(c\,x)^{m-j+1}\,\left(a\,x^j + b\,x^n\right)^{p+1}}{a\,(n-j)\,(p+1)}$$

Program code:

```
Int[(c_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  -c^(j-1)*(c*x)^(m-j+1)*(a*x^j+b*x^n)^(p+1)/(a*(n-j)*(p+1)) /;
FreeQ[{a,b,c,j,m,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && EqQ[m+n*p+n-j+1,0] && (IntegerQ[j] || GtQ[c,0])
```

3. $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p\, dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{m+n\,p+n-j+1}{n-j} \in \mathbb{Z}^-$

1: $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p\, dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{m+n\,p+n-j+1}{n-j} \in \mathbb{Z}^- \land p < -1 \land (j \in \mathbb{Z} \lor c > 0)$

Derivation: Generalized binomial recurrence 2b

Note: This rule increments $\frac{m+n\,p+n-j+1}{n-j}$ by 1 thus driving it to 0.

Rule: If $p \notin \mathbb{Z} \land j \neq n \land \frac{m+n\,p+n-j+1}{n-j} \in \mathbb{Z}^- \land p < -1 \land (j \in \mathbb{Z} \lor c > 0)$, then

$$\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p dx \longrightarrow -\frac{c^{j-1}\,(c\,x)^{m-j+1}\left(a\,x^j + b\,x^n\right)^{p+1}}{a\,(n-j)\,(p+1)} + \frac{c^j\,(m+n\,p+n-j+1)}{a\,(n-j)\,(p+1)} \int (c\,x)^{m-j} \left(a\,x^j + b\,x^n\right)^{p+1} dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  -c^(j-1)*(c*x)^(m-j+1)*(a*x^j+b*x^n)^(p+1)/(a*(n-j)*(p+1)) +
  c^j*(m+n*p+n-j+1)/(a*(n-j)*(p+1))*Int[(c*x)^(m-j)*(a*x^j+b*x^n)^(p+1),x] /;
FreeQ[{a,b,c,j,m,n},x] && Not[IntegerQ[p]] && NeQ[n,j] && ILtQ[Simplify[(m+n*p+n-j+1)/(n-j)],0] && LtQ[p,-1] && (IntegerQ[j] || GtQ[
```

2: $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p \,dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{m+n\,p+n-j+1}{n-j} \in \mathbb{Z}^- \land m + j\,p + 1 \neq 0 \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right)$

Derivation: Generalized binomial recurrence 3b

Note: This rule increments $\frac{m+n\,p+n-j+1}{n-j}$ by 1 thus driving it to 0.

Rule: If $p \notin \mathbb{Z} \land j \neq n \land \frac{m+n\,p+n-j+1}{n-j} \in \mathbb{Z}^- \land m + j\,p + 1 \neq 0 \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right)$, then

$$\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p \,dx \longrightarrow \frac{c^{j-1}\,(c\,x)^{m-j+1} \left(a\,x^j + b\,x^n\right)^{p+1}}{a\,(m + j\,p + 1)} - \frac{b\,(m + n\,p + n - j + 1)}{a\,c^{n-j}\,(m + j\,p + 1)} \int (c\,x)^{m+n-j} \left(a\,x^j + b\,x^n\right)^p \,dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  c^(j-1)*(c*x)^(m-j+1)*(a*x^j+b*x^n)^(p+1)/(a*(m+j*p+1)) -
  b*(m+n*p+n-j+1)/(a*c^(n-j)*(m+j*p+1))*Int[(c*x)^(m+n-j)*(a*x^j+b*x^n)^p,x] /;
FreeQ[{a,b,c,j,m,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && ILtQ[Simplify[(m+n*p+n-j+1)/(n-j)],0] && NeQ[m+j*p+1,0] && (IntegersQ[j,
```

3: $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p \, dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{m+n\,p+n-j+1}{n-j} \in \mathbb{Z}^- \land c \not> 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m} = 0$

Basis: $\frac{(c\,x)^m}{x^m} = \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule: If $p \notin \mathbb{Z} \land j \neq n \land \frac{m+n\,p+n-j+1}{n-j} \in \mathbb{Z}^- \land c \not> 0$, then

$$\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p \, dx \to \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a\,x^j + b\,x^n\right)^p \, dx$$

Program code:

```
Int[(c_*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a*x^j+b*x^n)^p,x] /;
FreeQ[{a,b,c,j,m,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && ILtQ[Simplify[(m+n*p+n-j+1)/(n-j)],0]
```

4. $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p \,dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land \frac{m+1}{n} \in \mathbb{Z} \land n^2 \neq 1$

1: $\int x^m \left(a\,x^j + b\,x^n\right)^p \,dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land \frac{m+1}{n} \in \mathbb{Z} \land n^2 \neq 1$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n} \in \mathbb{Z}$, then $x^m\,F[x^n] = \frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\,\partial_x x^n$

Note: If $n \in \mathbb{Z} \land \frac{m+1}{n} \in \mathbb{Z}$, then $m \in \mathbb{Z}$, and $(c\,x)^m$ automatically evaluates to $c^m\,x^m$.

Rule: If $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land \frac{m+1}{n} \in \mathbb{Z} \land n^2 \neq 1$, then

$$\int x^m \left(a\,x^j + b\,x^n\right)^p \,dx \,\longrightarrow\, \frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1} \left(a\,x^{j/n} + b\,x\right)^p \,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a*x^Simplify[j/n]+b*x)^p,x],x,x^n] /;
FreeQ[{a,b,j,m,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && IntegerQ[Simplify[j/n]] && IntegerQ[Simplify[(m+1)/n]] && NeQ[n^2,1]
```

2: $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p \,dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land \frac{m+1}{n} \in \mathbb{Z} \land n^2 \neq 1$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m} == 0$

Basis: $\frac{(c\,x)^m}{x^m} == \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule: If $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land \frac{m+1}{n} \in \mathbb{Z} \land n^2 \neq 1$, then

$$\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p \,dx \longrightarrow \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a\,x^j + b\,x^n\right)^p \,dx$$

Program code:

```
Int[(c_*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a*x^j+b*x^n)^p,x] /;
FreeQ[{a,b,c,j,m,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && IntegerQ[Simplify[j/n]] && IntegerQ[Simplify[(m+1)/n]] && NeQ[n^2,1]
```

5. $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right)$

   1. $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right) \land p > 0$

      1: $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right) \land p > 0 \land m + j\,p + 1 < 0$

Derivation: Generalized binomial recurrence 1a

Rule: If $p \notin \mathbb{Z} \land 0 < j < n \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right) \land p > 0 \land m + j\,p + 1 < 0$, then

$$\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p dx \longrightarrow \frac{(c\,x)^{m+1} \left(a\,x^j + b\,x^n\right)^p}{c\,(m + j\,p + 1)} - \frac{b\,p\,(n - j)}{c^n\,(m + j\,p + 1)} \int (c\,x)^{m+n} \left(a\,x^j + b\,x^n\right)^{p-1} dx$$

Program code:

```
Int[(c_.*x_)^m_*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a*x^j+b*x^n)^p/(c*(m+j*p+1)) -
  b*p*(n-j)/(c^n*(m+j*p+1))*Int[(c*x)^(m+n)*(a*x^j+b*x^n)^(p-1),x] /;
FreeQ[{a,b,c},x] && Not[IntegerQ[p]] && LtQ[0,j,n] && (IntegersQ[j,n] || GtQ[c,0]) && GtQ[p,0] && LtQ[m+j*p+1,0]
```

2: $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p\,dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right) \land p > 0 \land m + n\,p + 1 \neq 0$

Derivation: Generalized binomial recurrence 1b

Rule: If $p \notin \mathbb{Z} \land 0 < j < n \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right) \land p > 0 \land m + n\,p + 1 \neq 0$, then

$$\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p\,dx \longrightarrow \frac{(c\,x)^{m+1} \left(a\,x^j + b\,x^n\right)^p}{c\,(m + n\,p + 1)} + \frac{a\,p\,(n - j)}{c^j\,(m + n\,p + 1)} \int (c\,x)^{m+j} \left(a\,x^j + b\,x^n\right)^{p-1}\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a*x^j+b*x^n)^p/(c*(m+n*p+1)) +
  a*(n-j)*p/(c^j*(m+n*p+1))*Int[(c*x)^(m+j)*(a*x^j+b*x^n)^(p-1),x] /;
FreeQ[{a,b,c,m},x] && Not[IntegerQ[p]] && LtQ[0,j,n] && (IntegersQ[j,n] || GtQ[c,0]) && GtQ[p,0] && NeQ[m+n*p+1,0]
```

2. $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p \,dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right) \land p < -1$

1: $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p \,dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right) \land p < -1 \land m + j\,p + 1 > n - j$

Derivation: Generalized binomial recurrence 2a

Note: If $\frac{m+n\,p+n-j+1}{n-j} \in \mathbb{Z}^-$ following rule is used to drive $m + n\,p + n - j + 1$ to zero instead.

Rule: If $p \notin \mathbb{Z} \land 0 < j < n \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right) \land p < -1 \land m + j\,p + 1 > n - j$, then

$$\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p \,dx \to \frac{c^{n-1}\,(c\,x)^{m-n+1} \left(a\,x^j + b\,x^n\right)^{p+1}}{b\,(n-j)\,(p+1)} - \frac{c^n\,(m + j\,p - n + j + 1)}{b\,(n-j)\,(p+1)} \int (c\,x)^{m-n} \left(a\,x^j + b\,x^n\right)^{p+1} \,dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  c^(n-1)*(c*x)^(m-n+1)*(a*x^j+b*x^n)^(p+1)/(b*(n-j)*(p+1)) -
  c^n*(m+j*p-n+j+1)/(b*(n-j)*(p+1))*Int[(c*x)^(m-n)*(a*x^j+b*x^n)^(p+1),x] /;
FreeQ[{a,b,c},x] && Not[IntegerQ[p]] && LtQ[0,j,n] && (IntegersQ[j,n] || GtQ[c,0]) && LtQ[p,-1] && GtQ[m+j*p+1,n-j]
```

2: $\int (c\,x)^m \left(a\,x^j+b\,x^n\right)^p dx$ when $p \notin \mathbb{Z} \land 0<j<n \land \left((j \mid n) \in \mathbb{Z} \lor c>0\right) \land p<-1$

Derivation: Generalized binomial recurrence 2b

Rule: If $p \notin \mathbb{Z} \land 0<j<n \land \left((j \mid n) \in \mathbb{Z} \lor c>0\right) \land p<-1$, then

$$\int (c\,x)^m \left(a\,x^j+b\,x^n\right)^p dx \longrightarrow -\frac{c^{j-1}\,(c\,x)^{m-j+1}\left(a\,x^j+b\,x^n\right)^{p+1}}{a\,(n-j)\,(p+1)}+\frac{c^j\,(m+n\,p+n-j+1)}{a\,(n-j)\,(p+1)}\int (c\,x)^{m-j}\left(a\,x^j+b\,x^n\right)^{p+1} dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  -c^(j-1)*(c*x)^(m-j+1)*(a*x^j+b*x^n)^(p+1)/(a*(n-j)*(p+1)) +
  c^j*(m+n*p+n-j+1)/(a*(n-j)*(p+1))*Int[(c*x)^(m-j)*(a*x^j+b*x^n)^(p+1),x] /;
FreeQ[{a,b,c,m},x] && Not[IntegerQ[p]] && LtQ[0,j,n] && (IntegersQ[j,n] || GtQ[c,0]) && LtQ[p,-1]
```

3: $\int (c\,x)^m \left(a\,x^j+b\,x^n\right)^p dx$ when $p \notin \mathbb{Z} \land 0<j<n \land \left((j \mid n) \in \mathbb{Z} \lor c>0\right) \land m+j\,p+1>n-j \land m+n\,p+1 \neq 0$

Derivation: Generalized binomial recurrence 3a

Rule: If $p \notin \mathbb{Z} \land 0<j<n \land \left((j \mid n) \in \mathbb{Z} \lor c>0\right) \land m+j\,p+1>n-j \land m+n\,p+1 \neq 0$, then

$$\int (c\,x)^m \left(a\,x^j+b\,x^n\right)^p dx \longrightarrow \frac{c^{n-1}\,(c\,x)^{m-n+1}\left(a\,x^j+b\,x^n\right)^{p+1}}{b\,(m+n\,p+1)}-\frac{a\,c^{n-j}\,(m+j\,p-n+j+1)}{b\,(m+n\,p+1)}\int (c\,x)^{m-(n-j)}\left(a\,x^j+b\,x^n\right)^p dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  c^(n-1)*(c*x)^(m-n+1)*(a*x^j+b*x^n)^(p+1)/(b*(m+n*p+1)) -
  a*c^(n-j)*(m+j*p-n+j+1)/(b*(m+n*p+1))*Int[(c*x)^(m-(n-j))*(a*x^j+b*x^n)^p,x] /;
FreeQ[{a,b,c,m,p},x] && Not[IntegerQ[p]] && LtQ[0,j,n] && (IntegersQ[j,n] || GtQ[c,0]) && GtQ[m+j*p+1-n+j,0] && NeQ[m+n*p+1,0]
```

4: $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p\,dx$ when $p \notin \mathbb{Z} \land 0 < j < n \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right) \land m + j\,p + 1 < 0$

Derivation: Generalized binomial recurrence 3b

Rule: If $p \notin \mathbb{Z} \land 0 < j < n \land \left((j \mid n) \in \mathbb{Z} \lor c > 0\right) \land m + j\,p + 1 < 0$, then

$$\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p\,dx \longrightarrow \frac{c^{j-1}\,(c\,x)^{m-j+1}\left(a\,x^j + b\,x^n\right)^{p+1}}{a\left(m + j\,p + 1\right)} - \frac{b\left(m + n\,p + n - j + 1\right)}{a\,c^{n-j}\left(m + j\,p + 1\right)} \int (c\,x)^{m+n-j}\left(a\,x^j + b\,x^n\right)^p\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  c^(j-1)*(c*x)^(m-j+1)*(a*x^j+b*x^n)^(p+1)/(a*(m+j*p+1)) -
  b*(m+n*p+n-j+1)/(a*c^(n-j)*(m+j*p+1))*Int[(c*x)^(m+n-j)*(a*x^j+b*x^n)^p,x] /;
FreeQ[{a,b,c,m,p},x] && Not[IntegerQ[p]] && LtQ[0,j,n] && (IntegersQ[j,n] || GtQ[c,0]) && LtQ[m+j*p+1,0]
```

6. $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p \, dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land m+1 \neq 0 \land \frac{n}{m+1} \in \mathbb{Z}$

1: $\int x^m \left(a\,x^j + b\,x^n\right)^p \, dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land m+1 \neq 0 \land \frac{n}{m+1} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{n}{m+1} \in \mathbb{Z}$, then $x^m\,F[x^n] = \frac{1}{m+1}\,\text{Subst}\left[F\left[x^{\frac{n}{m+1}}\right], x, x^{m+1}\right] \partial_x x^{m+1}$

Rule: If $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land m+1 \neq 0 \land \frac{n}{m+1} \in \mathbb{Z}$, then

$$\int x^m \left(a\,x^j + b\,x^n\right)^p \, dx \longrightarrow \frac{1}{m+1}\,\text{Subst}\left[\int \left(a\,x^{\frac{j}{m+1}} + b\,x^{\frac{n}{m+1}}\right)^p dx,\ x,\ x^{m+1}\right]$$

Program code:

```
Int[x_^m_.*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  1/(m+1)*Subst[Int[(a*x^Simplify[j/(m+1)]+b*x^Simplify[n/(m+1)])^p,x],x,x^(m+1)] /;
FreeQ[{a,b,j,m,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && IntegerQ[Simplify[j/n]] && NeQ[m,-1] && IntegerQ[Simplify[n/(m+1)]] && Not
```

2: $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p\,dx$ when $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land m+1 \neq 0 \land \frac{n}{m+1} \in \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m} == 0$

Basis: $\frac{(c\,x)^m}{x^m} == \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule: If $p \notin \mathbb{Z} \land j \neq n \land \frac{j}{n} \in \mathbb{Z} \land m+1 \neq 0 \land \frac{n}{m+1} \in \mathbb{Z}$, then

$$\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p\,dx \longrightarrow \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a\,x^j + b\,x^n\right)^p\,dx$$

Program code:

```
Int[(c_*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a*x^j+b*x^n)^p,x] /;
FreeQ[{a,b,c,j,m,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && IntegerQ[Simplify[j/n]] && NeQ[m,-1] && IntegerQ[Simplify[n/(m+1)]] && N
```

7. $\int (c\,x)^m \left(a\,x^j+b\,x^n\right)^p dx$ when $p+\frac{1}{2}\in\mathbb{Z}\land j\neq n\land m+j\,p+1==0$

1. $\int (c\,x)^m \left(a\,x^j+b\,x^n\right)^p dx$ when $p+\frac{1}{2}\in\mathbb{Z}\land j\neq n\land m+j\,p+1==0\land (j\in\mathbb{Z}\lor c>0)$

1: $\int (c\,x)^m \left(a\,x^j+b\,x^n\right)^p dx$ when $p+\frac{1}{2}\in\mathbb{Z}^+\land j\neq n\land m+j\,p+1==0\land (j\in\mathbb{Z}\lor c>0)$

Derivation: Generalized binomial recurrence 1b

Rule: If $p+\frac{1}{2}\in\mathbb{Z}^+\land j\neq n\land m+j\,p+1==0\land (j\in\mathbb{Z}\lor c>0)$, then

$$\int (c\,x)^m \left(a\,x^j+b\,x^n\right)^p dx \to \frac{(c\,x)^{m+1}\left(a\,x^j+b\,x^n\right)^p}{c\,p\,(n-j)}+\frac{a}{c^j}\int (c\,x)^{m+j}\left(a\,x^j+b\,x^n\right)^{p-1}dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a*x^j+b*x^n)^p/(c*p*(n-j)) + a/c^j*Int[(c*x)^(m+j)*(a*x^j+b*x^n)^(p-1),x] /;
FreeQ[{a,b,c,j,m,n},x] && IGtQ[p+1/2,0] && NeQ[n,j] && EqQ[Simplify[m+j*p+1],0] && (IntegerQ[j] || GtQ[c,0])
```

2. $\int (c\,x)^m \left(a\,x^j+b\,x^n\right)^p dx$ when $p-\frac{1}{2}\in\mathbb{Z}^-\land j\neq n\land m+j\,p+1==0\land (j\in\mathbb{Z}\lor c>0)$

1: $\int \frac{x^m}{\sqrt{a\,x^j+b\,x^n}}\,dx$ when $m==\frac{j}{2}-1\land j\neq n$

Derivation: Integration by substitution

Basis: $\frac{x^{j/2-1}}{\sqrt{a\,x^j+b\,x^n}}==-\frac{2}{(n-j)}\,\text{Subst}\left[\frac{1}{1-a\,x^2},x,\frac{x^{j/2}}{\sqrt{a\,x^j+b\,x^n}}\right]\partial_x\frac{x^{j/2}}{\sqrt{a\,x^j+b\,x^n}}$

Rule: If $m==\frac{j}{2}-1\land j\neq n$, then

$$\int \frac{x^m}{\sqrt{a\,x^j + b\,x^n}}\,dx \longrightarrow -\frac{2}{(n-j)}\,\text{Subst}\left[\int \frac{1}{1-a\,x^2}\,dx,\ x,\ \frac{x^{j/2}}{\sqrt{a\,x^j + b\,x^n}}\right]$$

Program code:

```
Int[x_^m_./Sqrt[a_.*x_^j_.+b_.*x_^n_.],x_Symbol] :=
  -2/(n-j)*Subst[Int[1/(1-a*x^2),x],x,x^(j/2)/Sqrt[a*x^j+b*x^n]] /;
FreeQ[{a,b,j,n},x] && EqQ[m,j/2-1] && NeQ[n,j]
```

2: $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p dx$ when $p + \frac{1}{2} \in \mathbb{Z}^- \land j \neq n \land m + j\,p + 1 == 0 \land (j \in \mathbb{Z} \lor c > 0)$

Derivation: Generalized binomial recurrence 2b

Rule: If $p + \frac{1}{2} \in \mathbb{Z}^- \land j \neq n \land m + j\,p + 1 == 0 \land (j \in \mathbb{Z} \lor c > 0)$, then

$$\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p dx \longrightarrow -\frac{c^{j-1}\,(c\,x)^{m-j+1}\left(a\,x^j + b\,x^n\right)^{p+1}}{a\,(n-j)\,(p+1)} + \frac{c^j\,(m+n\,p+n-j+1)}{a\,(n-j)\,(p+1)} \int (c\,x)^{m-j}\left(a\,x^j + b\,x^n\right)^{p+1} dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  -c^(j-1)*(c*x)^(m-j+1)*(a*x^j+b*x^n)^(p+1)/(a*(n-j)*(p+1)) +
  c^j*(m+n*p+n-j+1)/(a*(n-j)*(p+1))*Int[(c*x)^(m-j)*(a*x^j+b*x^n)^(p+1),x] /;
FreeQ[{a,b,c,j,m,n},x] && ILtQ[p+1/2,0] && NeQ[n,j] && EqQ[Simplify[m+j*p+1],0] && (IntegerQ[j] || GtQ[c,0])
```

2: $\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p\,dx$ when $p + \frac{1}{2} \in \mathbb{Z} \land j \neq n \land m + j\,p + 1 == 0 \land \neg \left(j \in \mathbb{Z} \lor c > 0\right)$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m} == 0$

Basis: $\frac{(c\,x)^m}{x^m} == \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule: If $p + \frac{1}{2} \in \mathbb{Z} \land j \neq n \land m + j\,p + 1 == 0$, then

$$\int (c\,x)^m \left(a\,x^j + b\,x^n\right)^p\,dx \longrightarrow \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a\,x^j + b\,x^n\right)^p\,dx$$

Program code:

```
Int[(c_*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a*x^j+b*x^n)^p,x] /;
FreeQ[{a,b,c,j,m,n,p},x] && IntegerQ[p+1/2] && NeQ[n,j] && EqQ[Simplify[m+j*p+1],0]
```

x. $\int x^m \left(a x^j + b x^n\right)^p dx$ when $j \neq n$

1: $\int x^m \left(a x^j + b x^n\right)^p dx$ when $j \neq n \land m + j p + 1 == 0$

Note: Although this antiderivative appears simpler than that produced using piecewise constant extraction, *Mathematica* 8 has a hard time differentiating it back to $x^m \left(a x^j + b x^n\right)^p$.

Rule: If $j \neq n \land m + j p + 1 == 0$, then

$$\int x^m \left(a x^j + b x^n\right)^p dx \to \frac{\left(a x^j + b x^n\right)^{p+1}}{b p (n-j) x^{n+j p}} \text{Hypergeometric2F1}\left[1, 1, 1-p, -\frac{a}{b x^{n-j}}\right]$$

Program code:

```
(* Int[x_^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  (a*x^j+b*x^n)^(p+1)/(b*p*(n-j)*x^(n+j*p))*Hypergeometric2F1[1,1,1-p,-a/(b*x^(n-j))] /;
FreeQ[{a,b,j,m,n,p},x] && NeQ[n,j] && EqQ[m+j*p+1,0] *)
```

2: $\int x^m \left(a x^j + b x^n\right)^p dx$ when $j \neq n \land m + n + (p-1) j + 1 == 0$

Note: Although this antiderivative appears simpler than that produced using piecewise constant extraction, *Mathematica* 8 has a hard time differentiating it back to $x^m \left(a x^j + b x^n\right)^p$.

Rule: If $j \neq n \land m + n + (p-1) j + 1 == 0$, then

$$\int x^m \left(a x^j + b x^n\right)^p dx \to \frac{\left(a x^j + b x^n\right)^{p+1}}{b (p-1) (n-j) x^{2 n+j (p-1)}} \text{Hypergeometric2F1}\left[1, 2, 2-p, -\frac{a}{b x^{n-j}}\right]$$

Program code:

```
(* Int[x_^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  (a*x^j+b*x^n)^(p+1)/(b*(p-1)*(n-j)*x^(2*n+j*(p-1)))*Hypergeometric2F1[1,2,2-p,-a/(b*x^(n-j))] /;
FreeQ[{a,b,j,m,n,p},x] && NeQ[n,j] && EqQ[m+n+(p-1)*j+1,0] *)
```

3: $\int x^m \left(a x^j + b x^n\right)^p \, dx$ when $j \neq n \land m + j p + 1 \neq 0 \land m + n + (p - 1) j + 1 \neq 0$

Note: Although this antiderivative appears simpler than that produced using piecewise constant extraction, *Mathematica* 8 has a hard time differentiating it back to $x^m \left(a x^j + b x^n\right)^p$.

Rule: If $j \neq n \land m + j p + 1 \neq 0 \land m + n + (p - 1) j + 1 \neq 0$, then

$$\int x^m \left(a x^j + b x^n\right)^p \, dx \longrightarrow \frac{x^{m-j+1} \left(a x^j + b x^n\right)^{p+1}}{a \left(m + j p + 1\right)} \text{Hypergeometric2F1}\left[1, \frac{m + n p + 1}{n - j} + 1, \frac{m + j p + 1}{n - j} + 1, -\frac{b x^{n-j}}{a}\right]$$

Program code:

```
(* Int[x_^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  (x^(m-j+1)*(a*x^j+b*x^n)^(p+1))/(a*(m+j*p+1))*Hypergeometric2F1[1,(m+n*p+1)/(n-j)+1,(m+j*p+1)/(n-j)+1,-b*x^(n-j)/a] /;
FreeQ[{a,b,j,m,n,p},x] && NeQ[n,j] && NeQ[m+j*p+1,0] && NeQ[m+n+(p-1)*j+1,0] *)
```

8: $\int (c x)^m \left(a x^j + b x^n\right)^p \, dx$ when $p \notin \mathbb{Z} \land j \neq n$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c x)^m \left(a x^j + b x^n\right)^p}{x^{m+j p} \left(a + b x^{n-j}\right)^p} == 0$

Basis: $\frac{(c x)^m}{x^m} == \frac{c^{\text{IntPart}[m]} (c x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Basis: $\frac{\left(a x^j + b x^n\right)^p}{x^{j p} \left(a + b x^{n-j}\right)^p} == \frac{\left(a x^j + b x^n\right)^{\text{FracPart}[p]}}{x^{j \,\text{FracPart}[p]} \left(a + b x^{n-j}\right)^{\text{FracPart}[p]}}$

Rule: If $p \notin \mathbb{Z} \land j \neq n$, then

$$\int (c x)^m \left(a x^j + b x^n\right)^p \, dx \longrightarrow \frac{(c x)^m \left(a x^j + b x^n\right)^p}{x^{m+j p} \left(a + b x^{n-j}\right)^p} \int x^{m+j p} \left(a + b x^{n-j}\right)^p \, dx$$

$$\rightarrow \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}\,\left(a\,x^j+b\,x^n\right)^{\text{FracPart}[p]}}{x^{\text{FracPart}[m]+j\,\text{FracPart}[p]}\,\left(a+b\,x^{n-j}\right)^{\text{FracPart}[p]}}\int x^{m+j\,p}\,\left(a+b\,x^{n-j}\right)^p\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_.*x_^j_.+b_.*x_^n_.)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]*(a*x^j+b*x^n)^FracPart[p]/
    (x^(FracPart[m]+j*FracPart[p])*(a+b*x^(n-j))^FracPart[p])*
    Int[x^(m+j*p)*(a+b*x^(n-j))^p,x] /;
FreeQ[{a,b,c,j,m,n,p},x] && Not[IntegerQ[p]] && NeQ[n,j] && PosQ[n-j]
```

S: $\int u^m\,\left(a\,v^j+b\,v^n\right)^p\,dx$ when $v == c+d\,x \land u == e\,v$

Derivation: Integration by substitution and piecewise constant extraction

Basis: If $u == e\,v$, then $\partial_x \frac{u^m}{v^m} == 0$

Rule: If $v == c+d\,x \land u == e\,v$, then

$$\int u^m\,\left(a\,v^j+b\,v^n\right)^p\,dx \rightarrow \frac{u^m}{d\,v^m}\,\text{Subst}\left[\int x^m\,\left(a\,x^j+b\,x^n\right)^p\,dx,\,x,\,v\right]$$

Program code:

```
Int[u_^m_.*(a_.*v_^j_.+b_.*v_^n_.)^p_.,x_Symbol] :=
  u^m/(Coefficient[v,x,1]*v^m)*Subst[Int[x^m*(a*x^j+b*x^n)^p,x],x,v] /;
FreeQ[{a,b,j,m,n,p},x] && LinearPairQ[u,v,x]
```