Rules for integrands of the form $P_q[x]\ (a + b x^n + c x^{2n})^p$

1: $\int P_q[x]\ (a + b x^n + c x^{2n})^p\, dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int P_q[x]\ (a + b x^n + c x^{2n})^p\, dx \longrightarrow \int \text{ExpandIntegrand}\left[P_q[x]\ (a + b x^n + c x^{2n})^p,\ x\right] dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[Pq*(a+b*x^n+c*x^(2*n))^p,x],x] /;
FreeQ[{a,b,c,n},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && IGtQ[p,0]
```

2: $\int (d + e x^n + f x^{2n})\ (a + b x^n + c x^{2n})^p\, dx$ when $a e - b d\ (n\ (p + 1) + 1) == 0\ \land\ a f - c d\ (2 n\ (p + 1) + 1) == 0$

Rule: If $a e - b d\ (n\ (p + 1) + 1) == 0\ \land\ a f - c d\ (2 n\ (p + 1) + 1) == 0$, then

$$\int (d + e x^n + f x^{2n})\ (a + b x^n + c x^{2n})^p\, dx \longrightarrow \frac{d\, x\ (a + b x^n + c x^{2n})^{p+1}}{a}$$

Program code:

```
Int[(d_+e_.*x_^n_.+f_.*x_^n2_.)*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  d*x*(a+b*x^n+c*x^(2*n))^(p+1)/a /;
FreeQ[{a,b,c,d,e,f,n,p},x] && EqQ[n2,2*n] && EqQ[a*e-b*d*(n*(p+1)+1),0] && EqQ[a*f-c*d*(2*n*(p+1)+1),0]
```

```
Int[(d_+f_.*x_^n2_.)*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  d*x*(a+b*x^n+c*x^(2*n))^(p+1)/a /;
FreeQ[{a,b,c,d,f,n,p},x] && EqQ[n2,2*n] && EqQ[n*(p+1)+1,0] && EqQ[c*d+a*f,0]
```

3: $\int P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c=0\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c=0$, then $\partial_x\frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(b+2\,c\,x^n\right)^{2\,p}}=0$

Basis: If $b^2-4\,a\,c=0$, then $\frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(b+2\,c\,x^n\right)^{2\,p}}=\frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^{\text{FracPart}[p]}}{(4\,c)^{\text{IntPart}[p]}\,\left(b+2\,c\,x^n\right)^{2\,\text{FracPart}[p]}}$

Rule: If $b^2-4\,a\,c=0\,\land\,p\notin\mathbb{Z}$, then

$$\int P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\rightarrow\,\frac{\left(a+b\,x^n+c\,x^{2\,n}\right)^{\text{FracPart}[p]}}{(4\,c)^{\text{IntPart}[p]}\,\left(b+2\,c\,x^n\right)^{2\,\text{FracPart}[p]}}\int P_q[x]\,\left(b+2\,c\,x^n\right)^{2\,p}\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  (a+b*x^n+c*x^(2*n))^FracPart[p]/((4*c)^IntPart[p]*(b+2*c*x^n)^(2*FracPart[p]))*Int[Pq*(b+2*c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[2*p]]
```

4: $\int P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when **PolynomialRemainder[P_q[x], x, x] == 0**

Derivation: Algebraic simplification

Rule: If PolynomialRemainder[P_q[x], x, x] == 0, then

$$\int P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow\;\int x\,\text{PolynomialQuotient}[P_q[x],\,x,\,x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  Int[x*PolynomialQuotient[Pq,x,x]*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && EqQ[PolynomialRemainder[Pq,x,x],0] && Not[MatchQ[Pq,x^m_.*u_. /; IntegerQ[
```

5: $\int (d + e x^n + f x^{2n} + g x^{3n}) (a + b x^n + c x^{2n})^p \, dx$ when

$b^2 - 4 a c \neq 0 \land a^2 g (n + 1) - c (n (2 p + 3) + 1) (a e - b d (n (p + 1) + 1)) == 0 \land$
$a^2 f (n + 1) - a c d (n + 1) (2 n (p + 1) + 1) - b (n (p + 2) + 1) (a e - b d (n (p + 1) + 1)) == 0$

Rule: If
$b^2 - 4 a c \neq 0 \land a^2 g (n + 1) - c (n (2 p + 3) + 1) (a e - b d (n (p + 1) + 1)) == 0 \land$
$a^2 f (n + 1) - a c d (n + 1) (2 n (p + 1) + 1) - b (n (p + 2) + 1) (a e - b d (n (p + 1) + 1)) == 0$, then

$$\int (d + e x^n + f x^{2n} + g x^{3n}) (a + b x^n + c x^{2n})^p \, dx \to \frac{x (a d (n + 1) + (a e - b d (n (p + 1) + 1)) x^n) (a + b x^n + c x^{2n})^{p+1}}{a^2 (n + 1)}$$

Program code:

```
Int[(d_+e_.*x_^n_+f_.*x_^n2_.+g_.*x_^n3_.)*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  x*(a*d*(n+1)+(a*e-b*d*(n*(p+1)+1))*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/(a^2*(n+1)) /;
FreeQ[{a,b,c,d,e,f,g,n,p},x] && EqQ[n2,2*n] && EqQ[n3,3*n] && NeQ[b^2-4*a*c,0] &&
  EqQ[a^2*g*(n+1)-c*(n*(2*p+3)+1)*(a*e-b*d*(n*(p+1)+1)),0] &&
  EqQ[a^2*f*(n+1)-a*c*d*(n+1)*(2*n*(p+1)+1)-b*(n*(p+2)+1)*(a*e-b*d*(n*(p+1)+1)),0]
```

```
Int[(d_+f_.*x_^n2_.+g_.*x_^n3_.)*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  d*x*(a*(n+1)-b*(n*(p+1)+1)*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/(a^2*(n+1)) /;
FreeQ[{a,b,c,d,f,g,n,p},x] && EqQ[n2,2*n] && EqQ[n3,3*n] && NeQ[b^2-4*a*c,0] &&
  EqQ[a^2*g*(n+1)+c*b*d*(n*(2*p+3)+1)*(n*(p+1)+1),0] &&
  EqQ[a^2*f*(n+1)-a*c*d*(n+1)*(2*n*(p+1)+1)+b^2*d*(n*(p+2)+1)*(n*(p+1)+1),0]
```

```
Int[(d_+e_.*x_^n_+g_.*x_^n3_.)*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  x*(a*d*(n+1)+(a*e-b*d*(n*(p+1)+1))*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/(a^2*(n+1)) /;
FreeQ[{a,b,c,d,e,g,n,p},x] && EqQ[n2,2*n] && EqQ[n3,3*n] && NeQ[b^2-4*a*c,0] &&
  EqQ[a^2*g*(n+1)-c*(n*(2*p+3)+1)*(a*e-b*d*(n*(p+1)+1)),0] &&
  EqQ[a*c*d*(n+1)*(2*n*(p+1)+1)+b*(n*(p+2)+1)*(a*e-b*d*(n*(p+1)+1)),0]
```

```
Int[(d_+g_.*x_^n3_.)*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  d*x*(a*(n+1)-b*(n*(p+1)+1)*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/(a^2*(n+1)) /;
FreeQ[{a,b,c,d,g,n,p},x] && EqQ[n2,2*n] && EqQ[n3,3*n] && NeQ[b^2-4*a*c,0] &&
  EqQ[a^2*g*(n+1)+c*b*d*(n*(2*p+3)+1)*(n*(p+1)+1),0] &&
  EqQ[a*c*d*(n+1)*(2*n*(p+1)+1)-b^2*d*(n*(p+2)+1)*(n*(p+1)+1),0]
```

6. $\int P_q[x]\left(a+b x^n+c x^{2 n}\right)^p dx$ when $b^2-4 a c\neq 0\wedge n\in\mathbb{Z}$

1. $\int P_q[x]\left(a+b x^n+c x^{2 n}\right)^p dx$ when $b^2-4 a c\neq 0\wedge n\in\mathbb{Z}^+$

1. $\int P_q[x]\left(a+b x^n+c x^{2 n}\right)^p dx$ when $b^2-4 a c\neq 0\wedge n\in\mathbb{Z}^+\wedge p<-1$

1: $\int P_q[x]\left(a+b x^n+c x^{2 n}\right)^p dx$ when $b^2-4 a c\neq 0\wedge n\in\mathbb{Z}^+\wedge p<-1\wedge q<2 n$

Derivation: Trinomial recurrence 2b applied $n-1$ times

Rule: If $b^2-4 a c\neq 0\wedge n\in\mathbb{Z}^+\wedge p<-1\wedge q<2 n$, then

$$\int P_q[x]\left(a+b x^n+c x^{2 n}\right)^p dx\longrightarrow$$

$$-\frac{1}{a n (p+1)\left(b^2-4 a c\right)} x\left(a+b x^n+c x^{2 n}\right)^{p+1}\sum_{i=0}^{n-1}\left(\left(\left(b^2-2 a c\right) P_q[x,i]-a b P_q[x,n+i]\right) x^i+c\left(b P_q[x,i]-2 a P_q[x,n+i]\right) x^{n+i}\right)+$$
$$\frac{1}{a n (p+1)\left(b^2-4 a c\right)}\int\left(a+b x^n+c x^{2 n}\right)^{p+1}\cdot$$
$$\sum_{i=0}^{n-1}\left(\left(b^2 (n (p+1)+i+1)-2 a c (2 n (p+1)+i+1)\right) P_q[x,i]-a b (i+1) P_q[x,n+i]\right) x^i+c (n (2 p+3)+i+1)\left(b P_q[x,i]-2 a P_q[x,n+i]\right) x^{n+i}\right) dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  Module[{q=Expon[Pq,x],i},
  -x*(a+b*x^n+c*x^(2*n))^(p+1)/(a*n*(p+1)*(b^2-4*a*c))*
    Sum[((b^2-2*a*c)*Coeff[Pq,x,i]-a*b*Coeff[Pq,x,n+i])*x^i+
      c*(b*Coeff[Pq,x,i]-2*a*Coeff[Pq,x,n+i])*x^(n+i),{i,0,n-1}] +
  1/(a*n*(p+1)*(b^2-4*a*c))*Int[(a+b*x^n+c*x^(2*n))^(p+1)*
    Sum[((b^2*(n*(p+1)+i+1)-2*a*c*(2*n*(p+1)+i+1))*Coeff[Pq,x,i]-a*b*(i+1)*Coeff[Pq,x,n+i])*x^i+
      c*(n*(2*p+3)+i+1)*(b*Coeff[Pq,x,i]-2*a*Coeff[Pq,x,n+i])*x^(n+i),{i,0,n-1}],x] /;
 LtQ[q,2*n]] /;
FreeQ[{a,b,c},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1]
```

2: $\int P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,q\geq2\,n$

Derivation: Algebraic expansion and trinomial recurrence 2b applied $n-1$ times

Rule: If $b^2-4\,a\,c\neq0\,\wedge\,n\in\mathbb{Z}^+\wedge\,p<-1\,\wedge\,q\geq2\,n$, let $Q_{q-2\,n}[x]=\text{PolynomialQuotient}\left[P_q[x],\,a+b\,x^n+c\,x^{2\,n},\,x\right]$ and $R_{2\,n-1}[x]=\text{PolynomialRemainder}\left[P_q[x],\,a+b\,x^n+c\,x^{2\,n},\,x\right]$, then

$$\int P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow$$

$$\int R_{2\,n-1}[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx+\int Q_{q-2\,n}[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\,dx\,\longrightarrow$$

$$-\left(\left(x\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\sum_{i=0}^{n-1}\left(\left(\left(b^2-2\,a\,c\right)\,R_{2\,n-1}[x,\,i]-a\,b\,R_{2\,n-1}[x,\,n+i]\right)\,x^i+c\,\left(b\,R_{2\,n-1}[x,\,i]-2\,a\,R_{2\,n-1}[x,\,n+i]\right)\,x^{n+i}\right)\right)\Big/\left(a\,n\,(p+1)\,\left(b^2-4\,a\,c\right)\right)\right)+$$

$$\frac{1}{a\,n\,(p+1)\,\left(b^2-4\,a\,c\right)}\int\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\left(a\,n\,(p+1)\,\left(b^2-4\,a\,c\right)\,Q_{q-2\,n}[x]+\sum_{i=0}^{n-1}\left(\left(\left(b^2\,(n\,(p+1)+i+1)-2\,a\,c\,(2\,n\,(p+1)+i+1)\right)\,R_{2\,n-1}[x,\,i]-a\,b\,(i+1)\,R_{2\,n-1}[x,\,n+i]\right)\,x^i+c\,(n\,(2\,p+3)+i+1)\,\left(b\,R_{2\,n-1}[x,\,i]-2\,a\,R_{2\,n-1}[x,\,n+i]\right)\,x^{n+i}\right)\right)\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
  Module[{Q=PolynomialQuotient[(b*c)^(Floor[(q-1)/n]+1)*Pq,a+b*x^n+c*x^(2*n),x],
          R=PolynomialRemainder[(b*c)^(Floor[(q-1)/n]+1)*Pq,a+b*x^n+c*x^(2*n),x],i},
  -x*(a+b*x^n+c*x^(2*n))^(p+1)/(a*n*(p+1)*(b^2-4*a*c)*(b*c)^(Floor[(q-1)/n]+1))*
    Sum[((b^2-2*a*c)*Coeff[R,x,i]-a*b*Coeff[R,x,n+i])*x^i+
      c*(b*Coeff[R,x,i]-2*a*Coeff[R,x,n+i])*x^(n+i),{i,0,n-1}] +
   1/(a*n*(p+1)*(b^2-4*a*c)*(b*c)^(Floor[(q-1)/n]+1))*Int[(a+b*x^n+c*x^(2*n))^(p+1)*ExpandToSum[a*n*(p+1)*(b^2-4*a*c)*Q+
    Sum[((b^2*(n*(p+1)+i+1)-2*a*c*(2*n*(p+1)+i+1))*Coeff[R,x,i]-a*b*(i+1)*Coeff[R,x,n+i])*x^i+
      c*(n*(2*p+3)+i+1)*(b*Coeff[R,x,i]-2*a*Coeff[R,x,n+i])*x^(n+i),{i,0,n-1}],x],x]] /;
 GeQ[q,2*n]] /;
FreeQ[{a,b,c},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1]
```

2. $\int P_q[x^n]\ (a+b\,x^n+c\,x^{2\,n})^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+$

1: $\int\frac{P_q[x^n]}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ \text{NiceSqrtQ}[b^2-4\,a\,c]$

Derivation: Algebraic expansion

Rule: If $b^2-4\,a\,c\neq 0\ \wedge\ n\in\mathbb{Z}^+\ \wedge\ \text{NiceSqrtQ}[b^2-4\,a\,c]$, then

$$\int\frac{P_q[x^n]}{a+b\,x^n+c\,x^{2\,n}}\,dx\ \longrightarrow\ \int\text{ExpandIntegrand}\left[\frac{P_q[x^n]}{a+b\,x^n+c\,x^{2\,n}},\,x\right]dx$$

Program code:

```
Int[Pq_/(a_+b_.*x_^n_.+c_.*x_^n2_),x_Symbol] :=
  Int[ExpandIntegrand[Pq/(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && (NiceSqrtQ[b^2-4*a*c] || LtQ[Expon[Pq,x],n]
```

2. $\int P_q[x] \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0 \land 2\,p\in\mathbb{Z}^- \land q+2\,p+1==0$

1: $\int P_q[x] \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0 \land p\in\mathbb{Z}^- \land q+2\,p+1==0$

Note: This rule reduces the degree of the polynomial in the resulting integrand.

Rule: If $b^2-4\,a\,c\neq 0 \land p\in\mathbb{Z}^- \land q+2\,p+1==0$, then

$$\int P_q[x] \left(a+b\,x+c\,x^2\right)^p dx \longrightarrow$$
$$\frac{c^p\,Pq[x,q]\,Log\left[a+b\,x+c\,x^2\right]}{2}+\frac{1}{2}\int\left(2\,Pq[x]-\frac{c^p\,Pq[x,q]\,(b+2\,c\,x)}{\left(a+b\,x+c\,x^2\right)^{p+1}}\right)\left(a+b\,x+c\,x^2\right)^p dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    With[{Pqq=Coeff[Pq,x,q]},
    c^p*Pqq*Log[a+b*x+c*x^2]/2 +
    1/2*Int[ExpandToSum[2*Pq-c^p*Pqq*(b+2*c*x)/(a+b*x+c*x^2)^(p+1),x]*(a+b*x+c*x^2)^p,x]] /;
  EqQ[q+2*p+1,0]] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && ILtQ[p,0]
```

2. $\int P_q[x] \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0 \land 2\,p\in\mathbb{Z}^- \land q+2\,p+1==0$

1: $\int P_q[x] \left(a+b\,x+c\,x^2\right)^p dx$ when $b^2-4\,a\,c\neq 0 \land p+\frac{1}{2}\in\mathbb{Z}^- \land q+2\,p+1==0 \land c>0$

Note: This rule reduces the degree of the polynomial in the resulting integrand.

Rule: If $b^2-4\,a\,c\neq 0 \land p+\frac{1}{2}\in\mathbb{Z}^- \land q+2\,p+1==0 \land c>0$, then

$$\int P_q[x]\,\left(a+b\,x+c\,x^2\right)^p\,dx\;\longrightarrow$$

$$c^p\,Pq[x,\,q]\,\text{ArcTanh}\left[\frac{b+2\,c\,x}{2\,\sqrt{c}\,\sqrt{a+b\,x+c\,x^2}}\right]+\int\left(Pq[x]-\frac{c^{p+\frac{1}{2}}\,Pq[x,\,q]}{\left(a+b\,x+c\,x^2\right)^{p+\frac{1}{2}}}\right)\left(a+b\,x+c\,x^2\right)^p\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    With[{Pqq=Coeff[Pq,x,q]},
    c^p*Pqq*ArcTanh[(b+2*c*x)/(2*Rt[c,2]*Sqrt[a+b*x+c*x^2])] +
    Int[ExpandToSum[Pq-c^(p+1/2)*Pqq/(a+b*x+c*x^2)^(p+1/2),x]*(a+b*x+c*x^2)^p,x]] /;
  EqQ[q+2*p+1,0]] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && ILtQ[p+1/2,0] && PosQ[c]
```

2: $\int P_q[x]\left(a+b\,x+c\,x^2\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\ \wedge\ p+\frac{1}{2}\in\mathbb{Z}^-\ \wedge\ q+2\,p+1==0\ \wedge\ c\not>0$

Note: This rule reduces the degree of the polynomial in the resulting integrand.

Rule: If $b^2-4\,a\,c\neq 0\ \wedge\ p+\frac{1}{2}\in\mathbb{Z}^-\ \wedge\ q+2\,p+1==0\ \wedge\ c\not>0$, then

$$\int P_q[x]\left(a+b\,x+c\,x^2\right)^p\,dx\ \longrightarrow$$

$$-(-c)^p\,Pq[x,q]\,\text{ArcTan}\left[\frac{b+2\,c\,x}{2\,\sqrt{-c}\,\sqrt{a+b\,x+c\,x^2}}\right]+\int\left(Pq[x]-\frac{(-c)^{p+\frac{1}{2}}\,Pq[x,q]}{\left(a+b\,x+c\,x^2\right)^{p+\frac{1}{2}}}\right)\left(a+b\,x+c\,x^2\right)^p\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_+c_.*x_^2)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    With[{Pqq=Coeff[Pq,x,q]},
    -(-c)^p*Pqq*ArcTan[(b+2*c*x)/(2*Rt[-c,2]*Sqrt[a+b*x+c*x^2])] +
    Int[ExpandToSum[Pq-(-c)^(p+1/2)*Pqq/(a+b*x+c*x^2)^(p+1/2),x]*(a+b*x+c*x^2)^p,x]] /;
  EqQ[q+2*p+1,0]] /;
FreeQ[{a,b,c},x] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && ILtQ[p+1/2,0] && NegQ[c]
```

3: $\int P_q[x^n]\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+\land\,q\geq 2\,n\,\land\,q+2\,n\,p+1\neq 0$

Reference: G&R 2.160.3

Derivation: Trinomial recurrence 3a with $A=0$, $B=1$ and $m=m-n$

Reference: G&R 2.104

Note: This special case of the Ostrogradskiy-Hermite integration method reduces the degree of the polynomial in the resulting integrand.

Rule: If $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+\land\,q\geq 2\,n\,\land\,q+2\,n\,p+1\neq 0$, then

$$\int P_q[x^n]\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow$$

$$\int\left(P_q[x^n]-P_q[x,q]\,x^q\right)\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx+P_q[x,q]\int x^q\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow$$

$$\frac{P_q[x,q]\,x^{q-2\,n+1}\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}}{c\,(q+2\,n\,p+1)}+$$
$$\int\left(P_q[x^n]-P_q[x,q]\,x^q-\frac{P_q[x,q]\left(a\,(q-2\,n+1)\,x^{q-2\,n}+b\,(q+n\,(p-1)+1)\,x^{q-n}\right)}{c\,(q+2\,n\,p+1)}\right)\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    With[{Pqq=Coeff[Pq,x,q]},
    Pqq*x^(q-2*n+1)*(a+b*x^n+c*x^(2*n))^(p+1)/(c*(q+2*n*p+1)) +
    Int[ExpandToSum[Pq-Pqq*x^q-Pqq*(a*(q-2*n+1)*x^(q-2*n)+b*(q+n*(p-1)+1)*x^(q-n))/(c*(q+2*n*p+1)),x]*(a+b*x^n+c*x^(2*n))^p,x]]
  GeQ[q,2*n] && NeQ[q+2*n*p+1,0] && (IntegerQ[2*p] || EqQ[n,1] && IntegerQ[4*p] || IntegerQ[p+(q+1)/(2*n)])] /;
FreeQ[{a,b,c,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x^n] && NeQ[b^2-4*a*c,0] && IGtQ[n,0]
```

3: $\int P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+\land\,\neg\,\text{PolynomialQ}\left[P_q[x],\,x^n\right]$

Derivation: Algebraic expansion

Basis: If $n\in\mathbb{Z}^+$, then $P_q[x]=\sum_{j=0}^{n-1}x^j\sum_{k=0}^{(q-j)/n+1}P_q[x,\,j+k\,n]\,x^{k\,n}$

Note: This rule transform integrand into a sum of terms of the form $(d\,x)^k\,Q_r[x^n]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p$.

Rule: If $b^2-4\,a\,c\neq 0\,\land\,n\in\mathbb{Z}^+\land\,\neg\,\text{PolynomialQ}\left[P_q[x],\,x^n\right]$, then

$$\int P_q[x]\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\,\longrightarrow\,\int\sum_{j=0}^{n-1}x^j\left(\sum_{k=0}^{(q-j)/n+1}P_q[x,\,j+k\,n]\,x^{k\,n}\right)\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  Module[{q=Expon[Pq,x],j,k},
  Int[Sum[x^j*Sum[Coeff[Pq,x,j+k*n]*x^(k*n),{k,0,(q-j)/n+1}]*(a+b*x^n+c*x^(2*n))^p,{j,0,n-1}],x]] /;
FreeQ[{a,b,c,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && Not[PolyQ[Pq,x^n]]
```

4: $\int \frac{P_q[x]}{a+b\,x^n+c\,x^{2\,n}}\,dx$ when $b^2-4\,a\,c\neq 0\wedge n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $b^2-4\,a\,c\neq 0\wedge n\in\mathbb{Z}^+$, then

$$\int \frac{P_q[x]}{a+b\,x^n+c\,x^{2\,n}}\,dx\,\longrightarrow\,\int \text{RationalFunctionExpand}\left[\frac{P_q[x]}{a+b\,x^n+c\,x^{2\,n}},\,x\right]dx$$

Program code:

```
Int[Pq_/(a_+b_.*x_^n_.+c_.*x_^n2_.),x_Symbol] :=
  Int[RationalFunctionExpand[Pq/(a+b*x^n+c*x^(2*n)),x],x] /;
FreeQ[{a,b,c},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && IGtQ[n,0]
```

7: $\int P_q[x] \left(a+b x^n+c x^{2n}\right)^p \,dx$ when $b^2-4 a c \neq 0 \land n \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $g \in \mathbb{Z}^+$, then $P_q[x] \, F[x^n] = g \, \text{Subst}\left[x^{g-1} P_q[x^g] \, F[x^{g n}], x, x^{1/g}\right] \partial_x x^{1/g}$

Rule: If $b^2-4 a c \neq 0 \land n \in \mathbb{F}$, let $g = \text{Denominator}[n]$, then

$$\int P_q[x] \left(a+b x^n+c x^{2n}\right)^p \,dx \longrightarrow g \, \text{Subst}\left[\int x^{g-1} P_q[x^g] \left(a+b x^{g n}+c x^{2 g n}\right)^p \,dx, x, x^{1/g}\right]$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_,x_Symbol] :=
  With[{g=Denominator[n]},
  g*Subst[Int[x^(g-1)*ReplaceAll[Pq,x→x^g]*(a+b*x^(g*n)+c*x^(2*g*n))^p,x],x,x^(1/g)]] /;
FreeQ[{a,b,c,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0] && FractionQ[n]
```

8. $\int P_q[x]\left(a+b x^n+c x^{2n}\right)^p dx$ when $b^2-4ac\neq 0 \wedge p\in\mathbb{Z}^-$

1: $\int \frac{P_q[x]}{a+b x^n+c x^{2n}} dx$ when $b^2-4ac\neq 0$

Reference: G&R 2.161.1a

Derivation: Algebraic expansion

Basis: Let $q=\sqrt{b^2-4ac}$, then $\frac{1}{a+bz+cz^2}==\frac{2c}{q}\frac{1}{b-q+2cz}-\frac{2c}{q}\frac{1}{b+q+2cz}$

Rule: If $b^2-4ac\neq 0$, let $q=\sqrt{b^2-4ac}$, then

$$\int \frac{P_q[x]}{a+b x^n+c x^{2n}} dx \longrightarrow \frac{2c}{q}\int \frac{P_q[x]}{b-q+2c x^n} dx-\frac{2c}{q}\int \frac{P_q[x]}{b+q+2c x^n} dx$$

Program code:

```
Int[Pq_/(a_+b_.*x_^n_.+c_.*x_^n2_.),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[Pq/(b-q+2*c*x^n),x] -
  2*c/q*Int[Pq/(b+q+2*c*x^n),x]] /;
FreeQ[{a,b,c,n},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && NeQ[b^2-4*a*c,0]
```

?: $\int\left(A+B x^n+C x^{2n}+D x^{3n}\right)\left(a+b x^n+c x^{2n}\right)^p dx$ when $b^2-4ac\neq 0 \wedge p+1\in\mathbb{Z}^-$

Derivation: Two steps of OS and trinomial recurrence 2b

Note: This rule should be generalized for integrands of the form $P_q[x^n]\left(a+b x^n+c x^{2n}\right)^p$ when n is symbolic.

Rule 1.3.3.17: If $b^2-4ac\neq 0 \wedge p+1\in\mathbb{Z}^-$, then

$$\int\left(d+e\,x^n+f\,x^{2\,n}+g\,x^{3\,n}\right)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^p\,dx\;\longrightarrow$$

$$-\left(\left(x\,\left(b^2\,c\,d-2\,a\,c\,\left(c\,d-a\,f\right)-a\,b\,(c\,e+a\,g)+\left(b\,c\,\left(c\,d+a\,f\right)-a\,b^2\,g-2\,a\,c\,(c\,e-a\,g)\right)\,x^n\right)\,\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\right)\Big/\left(a\,c\,n\,(p+1)\,\left(b^2-4\,a\,c\right)\right)\right)-$$
$$\frac{1}{a\,c\,n\,(p+1)\,\left(b^2-4\,a\,c\right)}\int\left(a+b\,x^n+c\,x^{2\,n}\right)^{p+1}\,\left(a\,b\,(c\,e+a\,g)-b^2\,c\,d\,(n+n\,p+1)-2\,a\,c\,\left(a\,f-c\,d\,(2\,n\,(p+1)+1)\right)+\right.$$
$$\left.\left(a\,b^2\,g\,(n\,(p+2)+1)-b\,c\,\left(c\,d+a\,f\right)\,(n\,(3+2\,p)+1)-2\,a\,c\,(a\,g\,(n+1)-c\,e\,(n\,(2\,p+3)+1))\right)\,x^n\right)\,dx$$

Program code:

```
Int[P3_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  With[{d=Coeff[P3,x^n,0],e=Coeff[P3,x^n,1],f=Coeff[P3,x^n,2],g=Coeff[P3,x^n,3]},
  -x*(b^2*c*d-2*a*c*(c*d-a*f)-a*b*(c*e+a*g)+(b*c*(c*d+a*f)-a*b^2*g-2*a*c*(c*e-a*g))*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/
    (a*c*n*(p+1)*(b^2-4*a*c)) -
  1/(a*c*n*(p+1)*(b^2-4*a*c))*Int[(a+b*x^n+c*x^(2*n))^(p+1)*
     Simp[a*b*(c*e+a*g)-b^2*c*d*(n+n*p+1)-2*a*c*(a*f-c*d*(2*n*(p+1)+1))+
        (a*b^2*g*(n*(p+2)+1)-b*c*(c*d+a*f)*(n*(2*p+3)+1)-2*a*c*(a*g*(n+1)-c*e*(n*(2*p+3)+1)))*x^n,x],x]] /;
FreeQ[{a,b,c,n},x] && EqQ[n2,2*n] && PolyQ[P3,x^n,3] && NeQ[b^2-4*a*c,0] && ILtQ[p,-1]
```

```
Int[P2_*(a_+b_.*x_^n_+c_.*x_^n2_)^p_,x_Symbol] :=
  With[{d=Coeff[P2,x^n,0],e=Coeff[P2,x^n,1],f=Coeff[P2,x^n,2]},
  -x*(b^2*d-2*a*(c*d-a*f)-a*b*e+(b*(c*d+a*f)-2*a*c*e)*x^n)*(a+b*x^n+c*x^(2*n))^(p+1)/(a*n*(p+1)*(b^2-4*a*c)) -
  1/(a*n*(p+1)*(b^2-4*a*c))*Int[(a+b*x^n+c*x^(2*n))^(p+1)*
     Simp[a*b*e-b^2*d*(n+n*p+1)-2*a*(a*f-c*d*(2*n*(p+1)+1))-(b*(c*d+a*f)*(n*(2*p+3)+1)-2*a*c*e*(n*(2*p+3)+1))*x^n,x],x]] /;
FreeQ[{a,b,c,n},x] && EqQ[n2,2*n] && PolyQ[P2,x^n,2] && NeQ[b^2-4*a*c,0] && ILtQ[p,-1]
```

2: $\int P_q[x]\left(a+b x^n+c x^{2 n}\right)^p dx$ when $p+1 \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Rule: If $p+1 \in \mathbb{Z}^-$, then

$$\int P_q[x]\left(a+b x^n+c x^{2 n}\right)^p dx \to \int \text{ExpandIntegrand}\left[P_q[x]\left(a+b x^n+c x^{2 n}\right)^p, x\right] dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_,x_Symbol] :=
  Int[ExpandIntegrand[Pq*(a+b*x^n+c*x^(2*n))^p,x],x] /;
FreeQ[{a,b,c,n},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && ILtQ[p,-1]
```

X: $\int P_q[x]\left(a+b x^n+c x^{2 n}\right)^p dx$

Rule:

$$\int P_q[x]\left(a+b x^n+c x^{2 n}\right)^p dx \to \int P_q[x]\left(a+b x^n+c x^{2 n}\right)^p dx$$

Program code:

```
Int[Pq_*(a_+b_.*x_^n_.+c_.*x_^n2_.)^p_.,x_Symbol] :=
  Unintegrable[Pq*(a+b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[n2,2*n] && (PolyQ[Pq,x] || PolyQ[Pq,x^n])
```

S: $\int P_q[v^n]\ (a+b\,v^n+c\,v^{2\,n})^p\, dx$ when $v == f+g\,x$

Derivation: Integration by substitution

Rule: If $v == f+g\,x$, then

$$\int P_q[v^n]\ (a+b\,v^n+c\,v^{2\,n})^p\, dx \longrightarrow \frac{1}{g}\,\mathrm{Subst}\left[\int P_q[x^n]\ (a+b\,x^n+c\,x^{2\,n})^p\, dx,\ x,\ v\right]$$

Program code:

```
Int[Pq_*(a_+b_.*v_^n_+c_.*v_^n2_.)^p_.,x_Symbol] :=
  1/Coefficient[v,x,1]*Subst[Int[SubstFor[v,Pq,x]*(a+b*x^n+c*x^(2*n))^p,x],x,v] /;
FreeQ[{a,b,c,n,p},x] && EqQ[n2,2*n] && LinearQ[v,x] && PolyQ[Pq,v^n]
```