?. $\int \frac{(d\,x)^m\left(e+f\,x^{n/4}+g\,x^{3\,n/4}+h\,x^n\right)}{\left(a+c\,x^n\right)^{3/2}}\,dx$ when $4\,m-n+4==0\,\wedge\,c\,e+a\,h==0$

1: $\int \frac{x^m\left(e+f\,x^{n/4}+g\,x^{3\,n/4}+h\,x^n\right)}{\left(a+c\,x^n\right)^{3/2}}\,dx$ when $4\,m-n+4==0\,\wedge\,c\,e+a\,h==0$

Rule: If $4\,m-n+4==0\,\wedge\,c\,e+a\,h==0$, then

$$\int \frac{x^m\left(e+f\,x^{n/4}+g\,x^{3\,n/4}+h\,x^n\right)}{\left(a+c\,x^n\right)^{3/2}}\,dx\,\longrightarrow\,-\frac{2\,a\,g+4\,a\,h\,x^{n/4}-2\,c\,f\,x^{n/2}}{a\,c\,n\,\sqrt{a+c\,x^n}}$$

Program code:

```
Int[x_^m_.*(e_+f_.*x_^q_.+g_.*x_^r_.+h_.*x_^n_.)/(a_+c_.*x_^n_.)^(3/2),x_Symbol] :=
  -(2*a*g+4*a*h*x^(n/4)-2*c*f*x^(n/2))/(a*c*n*Sqrt[a+c*x^n]) /;
FreeQ[{a,c,e,f,g,h,m,n},x] && EqQ[q,n/4] && EqQ[r,3*n/4] && EqQ[4*m-n+4,0] && EqQ[c*e+a*h,0]
```

2: $\int \frac{(d\,x)^m\left(e+f\,x^{n/4}+g\,x^{3\,n/4}+h\,x^n\right)}{\left(a+c\,x^n\right)^{3/2}}\,dx$ when $4\,m-n+4==0\,\wedge\,c\,e+a\,h==0$

Rule: If $4\,m-n+4==0\,\wedge\,c\,e+a\,h==0$, then

$$\int \frac{(d\,x)^m\left(e+f\,x^{n/4}+g\,x^{3\,n/4}+h\,x^n\right)}{\left(a+c\,x^n\right)^{3/2}}\,dx\,\longrightarrow\,\frac{(d\,x)^m}{x^m}\int \frac{x^m\left(e+f\,x^{n/4}+g\,x^{3\,n/4}+h\,x^n\right)}{\left(a+c\,x^n\right)^{3/2}}\,dx$$

Program code:

```
Int[(d_*x_)^m_.*(e_+f_.*x_^q_.+g_.*x_^r_.+h_.*x_^n_.)/(a_+c_.*x_^n_.)^(3/2),x_Symbol] :=
  (d*x)^m/x^m*Int[x^m*(e+f*x^(n/4)+g*x^((3*n)/4)+h*x^n)/(a+c*x^n)^(3/2),x] /;
FreeQ[{a,c,d,e,f,g,h,m,n},x] && EqQ[4*m-n+4,0] && EqQ[q,n/4] && EqQ[r,3*n/4] && EqQ[c*e+a*h,0]
```

Rules for integrands of the form $(c\,x)^m\,P_q[x]\,(a+b\,x^n)^p$

1: $\int (c\,x)^m\,P_q[x]\,(a+b\,x)^p\,dx$ when $p\in\mathbb{F}\,\wedge\,m+1\in\mathbb{Z}^-$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}^+$, then $F[x]\,(a+b\,x)^p == \frac{n}{b}\,\text{Subst}\left[x^{n\,p+n-1}\,F\left[-\frac{a}{b}+\frac{x^n}{b}\right],\,x,\,(a+b\,x)^{1/n}\right]\,\partial_x\,(a+b\,x)^{1/n}$

Rule: If $p\in\mathbb{F}\,\wedge\,m+1\in\mathbb{Z}^-$, let $n=\text{Denominator}[p]$, then

$$\int (c\,x)^m\,P_q[x]\,(a+b\,x)^p\,dx\;\longrightarrow\;\frac{n}{b}\,\text{Subst}\left[\int x^{n\,p+n-1}\left(-\frac{a\,c}{b}+\frac{c\,x^n}{b}\right)^m P_q\left[-\frac{a}{b}+\frac{x^n}{b}\right]dx,\,x,\,(a+b\,x)^{1/n}\right]$$

Program code:

```
Int[(c_.*x_)^m_*Pq_*(a_+b_.*x_)^p_,x_Symbol] :=
  With[{n=Denominator[p]},
  n/b*Subst[Int[x^(n*p+n-1)*(-a*c/b+c*x^n/b)^m*ReplaceAll[Pq,x→-a/b+x^n/b],x],x,(a+b*x)^(1/n)]] /;
FreeQ[{a,b,c,m},x] && PolyQ[Pq,x] && FractionQ[p] && ILtQ[m,-1]
```

2: $\int x^m P_q\left[x^{m+1}\right] \left(a+b x^n\right)^p dx$ when $m \neq -1 \land \frac{n}{m+1} \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: $x^m F\left[x^{m+1}\right] == \frac{1}{m+1} \text{Subst}\left[F[x], x, x^{m+1}\right] \partial_x x^{m+1}$

Rule: If $m \neq -1 \land \frac{n}{m+1} \in \mathbb{Z}^+$, then

$$\int x^m P_q\left[x^{m+1}\right] \left(a+b x^n\right)^p dx \rightarrow \frac{1}{m+1} \text{Subst}\left[\int P_q[x] \left(a+b x^{\frac{n}{m+1}}\right)^p dx, x, x^{m+1}\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_)^p_.,x_Symbol] :=
  1/(m+1)*Subst[Int[SubstFor[x^(m+1),Pq,x]*(a+b*x^Simplify[n/(m+1)])^p,x],x,x^(m+1)] /;
FreeQ[{a,b,m,n,p},x] && NeQ[m,-1] && IGtQ[Simplify[n/(m+1)],0] && PolyQ[Pq,x^(m+1)]
```

3. $\int (c\,x)^m\, P_q[x]\, \left(a + b\,x^n\right)^p\, dx$ when $p \in \mathbb{Z}^+$

1: $\int x^m\, P_q[x]\, \left(a + b\,x^n\right)^p\, dx$ when $p \in \mathbb{Z}^+ \land n - m \in \mathbb{Z}^+ \land P_q[x,\, n - m - 1] \neq 0$

Derivation: Algebraic expansion

Basis: $\int x^{n-1}\, (a + b\,x^n)^p\, dx == \frac{(a+b\,x^n)^{p+1}}{b\,n\,(p+1)}$

Rule: If $p \in \mathbb{Z}^+ \land n - m \in \mathbb{Z}^+ \land P_q[x,\, n - m - 1] \neq 0$, then

$$\int x^m\, P_q[x]\, \left(a + b\,x^n\right)^p\, dx \longrightarrow \frac{P_q[x,\, n - m - 1]\, \left(a + b\,x^n\right)^{p+1}}{b\,n\,(p + 1)} + \int x^m\, \left(P_q[x] - P_q[x,\, n - m - 1]\, x^{n-m-1}\right)\, \left(a + b\,x^n\right)^p\, dx$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_.)^p_,x_Symbol] :=
  Coeff[Pq,x,n-m-1]*(a+b*x^n)^(p+1)/(b*n*(p+1)) +
  Int[x^m*ExpandToSum[Pq-Coeff[Pq,x,n-m-1]*x^(n-m-1),x]*(a+b*x^n)^p,x] /;
FreeQ[{a,b,m,n},x] && PolyQ[Pq,x] && IGtQ[p,0] && IGtQ[n-m,0] && NeQ[Coeff[Pq,x,n-m-1],0]
```

2: $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $p \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+$, then

$$\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx \longrightarrow \int \text{ExpandIntegrand}\left[(c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p,\,x\right]\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^n_.)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(c*x)^m*Pq*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,m,n},x] && PolyQ[Pq,x] && (IGtQ[p,0] || EqQ[n,1])
```

4. $\int (c\,x)^m\,P_q[x^n]\,(a+b\,x^n)^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

1: $\int x^m\,P_q[x^n]\,(a+b\,x^n)^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n}\in\mathbb{Z}$, then $x^m\,F[x^n] = \frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\,\partial_x x^n$

Note: If $n\in\mathbb{Z}\wedge\frac{m+1}{n}\in\mathbb{Z}$, then $m\in\mathbb{Z}$, and $(c\,x)^m$ automatically evaluates to $c^m\,x^m$.

Rule: If $\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int x^m\,P_q[x^n]\,(a+b\,x^n)^p\,dx\;\longrightarrow\;\frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1}\,P_q[x]\,(a+b\,x)^p\,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_)^p_.,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*SubstFor[x^n,Pq,x]*(a+b*x)^p,x],x,x^n] /;
FreeQ[{a,b,m,n,p},x] && PolyQ[Pq,x^n] && IntegerQ[Simplify[(m+1)/n]]
```

2: $\int (c\,x)^m\,P_q[x^n]\,(a+b\,x^n)^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m} == 0$

Basis: $\frac{(c\,x)^m}{x^m} == \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule: If $\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int (c\,x)^m\,P_q[x^n]\,(a+b\,x^n)^p\,dx \longrightarrow \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,P_q[x^n]\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[(c_*x_)^m_.*Pq_*(a_+b_.*x_^n_)^p_.,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*Pq*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,m,n,p},x] && PolyQ[Pq,x^n] && IntegerQ[Simplify[(m+1)/n]]
```

5: $\int x^m P_q[x] \left(a+b x^n\right)^p dx$ when $m-n+1=0 \land p<-1$

Derivation: Integration by parts

Basis: $x^{n-1} (a+b x^n)^p = \partial_x \frac{(a+b x^n)^{p+1}}{b n (p+1)}$

Rule: If $m-n+1=0 \land p<-1$, then

$$\int x^m P_q[x] \left(a+b x^n\right)^p dx \rightarrow \frac{P_q[x] \left(a+b x^n\right)^{p+1}}{b n (p+1)} - \frac{1}{b n (p+1)} \int \partial_x P_q[x] \left(a+b x^n\right)^{p+1} dx$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  Pq*(a+b*x^n)^(p+1)/(b*n*(p+1)) -
  1/(b*n*(p+1))*Int[D[Pq,x]*(a+b*x^n)^(p+1),x] /;
FreeQ[{a,b,m,n},x] && PolyQ[Pq,x] && EqQ[m-n+1,0] && LtQ[p,-1]
```

6: $\int (d\,x)^m\,P_q[x]\,(a+b\,x^n)^p\,dx$ when $\text{PolynomialRemainder}[P_q[x],\,x,\,x]=0$

Derivation: Algebraic simplification

Rule: If $\text{PolynomialRemainder}[P_q[x],\,x,\,x]=0$, then

$$\int (d\,x)^m\,P_q[x]\,(a+b\,x^n)^p\,dx \rightarrow \frac{1}{d}\int (d\,x)^{m+1}\,\text{PolynomialQuotient}[P_q[x],\,x,\,x]\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[(d_.*x_)^m_.*Pq_*(a_+b_.*x_^n_.)^p_,x_Symbol] :=
  1/d*Int[(d*x)^(m+1)*PolynomialQuotient[Pq,x,x]*(a+b*x^n)^p,x] /;
FreeQ[{a,b,d,m,n,p},x] && PolyQ[Pq,x] && EqQ[PolynomialRemainder[Pq,x,x],0]
```

7. $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{Z}$

1. $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{Z}^+$

1. $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{Z}^+\land p>0$

1: $\int x^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{Z}^+\land p>0\land m+q+1<0$

Derivation: Integration by parts

Rule: If $n\in\mathbb{Z}^+\land\ p>0\ \land\ m+q+1<0$, let $u=\int x^m\,P_q[x]\,dx$ then

$$\int x^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx\;\longrightarrow\;u\,\left(a+b\,x^n\right)^p-b\,n\,p\int x^{m+n}\,\left(a+b\,x^n\right)^{p-1}\,\frac{u}{x^{m+1}}\,dx$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_.)^p_,x_Symbol] :=
  Module[{u=IntHide[x^m*Pq,x]},
  u*(a+b*x^n)^p - b*n*p*Int[x^(m+n)*(a+b*x^n)^(p-1)*ExpandToSum[u/x^(m+1),x],x]] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[n,0] && GtQ[p,0] && LtQ[m+Expon[Pq,x]+1,0]
```

2: $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $\frac{n-1}{2}\in\mathbb{Z}^+\wedge\,p>0$

Derivation: Binomial recurrence 1b applied q times

Rule: If $\frac{n-1}{2}\in\mathbb{Z}^+\wedge\,p>0$, then

$$\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx\;\longrightarrow\;(c\,x)^m\,\left(a+b\,x^n\right)^p\sum_{i=0}^{q}\frac{P_q[x,\,i]\,x^{i+1}}{m+n\,p+i+1}+a\,n\,p\int (c\,x)^m\,\left(a+b\,x^n\right)^{p-1}\left(\sum_{i=0}^{q}\frac{P_q[x,\,i]\,x^{i}}{m+n\,p+i+1}\right)dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^n_.)^p_,x_Symbol] :=
  Module[{q=Expon[Pq,x],i},
  (c*x)^m*(a+b*x^n)^p*Sum[Coeff[Pq,x,i]*x^(i+1)/(m+n*p+i+1),{i,0,q}] +
   a*n*p*Int[(c*x)^m*(a+b*x^n)^(p-1)*Sum[Coeff[Pq,x,i]*x^i/(m+n*p+i+1),{i,0,q}],x]] /;
FreeQ[{a,b,c,m},x] && PolyQ[Pq,x] && IGtQ[(n-1)/2,0] && GtQ[p,0]
```

2. $\int x^m P_q[x] \left(a+b x^n\right)^p dx$ when $n \in \mathbb{Z}^+ \wedge p < -1 \wedge m \in \mathbb{Z}$

1. $\int x^m P_q[x] \left(a+b x^n\right)^p dx$ when $n \in \mathbb{Z}^+ \wedge p < -1 \wedge m \in \mathbb{Z}^+$

1: $\int \frac{x^2 \left(e+f x+h x^4\right)}{\left(a+b x^4\right)^{3/2}} dx$ when $b e-3 a h == 0$

Rule: If $b e-3 a h == 0$, then

$$\int \frac{x^2 \left(e+f x+h x^4\right)}{\left(a+b x^4\right)^{3/2}} dx \longrightarrow -\frac{f-2 h x^3}{2 b \sqrt{a+b x^4}}$$

Program code:

```
Int[x_^2*P4_/(a_+b_.*x_^4)^(3/2),x_Symbol] :=
  With[{e=Coeff[P4,x,0],f=Coeff[P4,x,1],h=Coeff[P4,x,4]},
  -(f-2*h*x^3)/(2*b*Sqrt[a+b*x^4]) /;
 EqQ[b*e-3*a*h,0]] /;
FreeQ[{a,b},x] && PolyQ[P4,x,4] && EqQ[Coeff[P4,x,2],0] && EqQ[Coeff[P4,x,3],0]
```

2: $\int x^m P_q[x] \left(a+b x^n\right)^p dx$ when $n \in \mathbb{Z}^+ \wedge p < -1 \wedge m \in \mathbb{Z}^+ \wedge m+q \geq n$

Derivation: Algebraic expansion and binomial recurrence 2b applied $n-1$ times

Note: $\sum_{i=0}^{q} (i+1) P_q[x, i] x^i == \partial_x (x P_q[x])$ contributed by Martin Welz on 5 June 2015

Rule: If $n \in \mathbb{Z}^+ \wedge p < -1 \wedge m \in \mathbb{Z}^+ \wedge m+q \geq n$, let $Q_{m+q-n}[x] \rightarrow \text{PolynomialQuotient}\left[x^m P_q[x], a+b x^n, x\right]$ and $R_{n-1}[x] \rightarrow \text{PolynomialRemainder}\left[x^m P_q[x], a+b x^n, x\right]$, then

$$\int x^m P_q[x] \left(a+b x^n\right)^p dx \longrightarrow$$

$$\int R_{n-1}[x] \left(a+b x^n\right)^p dx+\int Q_{m+q-n}[x] \left(a+b x^n\right)^{p+1} dx \longrightarrow$$

$$-\frac{x R_{n-1}[x] \left(a+b x^n\right)^{p+1}}{a n (p+1)}+\frac{1}{a n (p+1)} \int (a n (p+1) Q_{m+q-n}[x]+n (p+1) R_{n-1}[x]+\partial_x (x R_{n-1}[x])) \left(a+b x^n\right)^{p+1} dx$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_.)^p_,x_Symbol] :=
  With[{q=m+Expon[Pq,x]},
    Module[{Q=PolynomialQuotient[b^(Floor[(q-1)/n]+1)*x^m*Pq,a+b*x^n,x],
            R=PolynomialRemainder[b^(Floor[(q-1)/n]+1)*x^m*Pq,a+b*x^n,x]},
    -x*R*(a+b*x^n)^(p+1)/(a*n*(p+1)*b^(Floor[(q-1)/n]+1)) +
     1/(a*n*(p+1)*b^(Floor[(q-1)/n]+1))*Int[(a+b*x^n)^(p+1)*ExpandToSum[a*n*(p+1)*Q+n*(p+1)*R+D[x*R,x],x],x]] /;
  GeQ[q,n]] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[n,0] && LtQ[p,-1] && IGtQ[m,0]
```

2: $\int x^m P_q[x] \left(a+b x^n\right)^p \, dx$ when $n \in \mathbb{Z}^+ \land p < -1 \land m \in \mathbb{Z}^-$

Derivation: Algebraic expansion and binomial recurrence 2b applied $n-1$ times

Rule: If $n \in \mathbb{Z}^+ \land p < -1 \land m \in \mathbb{Z}^-$, let $Q_{q-n}[x] = \text{PolynomialQuotient}\left[x^m P_q[x], a+b x^n, x\right]$ and $R_{n-1}[x] = \text{PolynomialRemainder}\left[x^m P_q[x], a+b x^n, x\right]$, then

$$\int x^m P_q[x] \left(a+b x^n\right)^p \, dx \longrightarrow$$

$$\int R_{n-1}[x] \left(a+b x^n\right)^p \, dx + \int Q_{q-n}[x] \left(a+b x^n\right)^{p+1} \, dx \longrightarrow$$

$$-\frac{x\, R_{n-1}[x] \left(a+b x^n\right)^{p+1}}{a\, n\, (p+1)} + \frac{1}{a\, n\, (p+1)} \int x^m \left(a\, n\, (p+1)\, x^{-m}\, Q_{q-n}[x] + \sum_{i=0}^{n-1} (n\, (p+1) + i + 1)\, R_{n-1}[x, i]\, x^{i-m}\right) \left(a+b x^n\right)^{p+1} \, dx$$

Program code:

```
Int[x_^m_*Pq_*(a_+b_.*x_^n_.)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
  Module[{Q=PolynomialQuotient[a*b^(Floor[(q-1)/n]+1)*x^m*Pq,a+b*x^n,x],
          R=PolynomialRemainder[a*b^(Floor[(q-1)/n]+1)*x^m*Pq,a+b*x^n,x],i},
    -x*R*(a+b*x^n)^(p+1)/(a^2*n*(p+1)*b^(Floor[(q-1)/n]+1)) +
    1/(a*n*(p+1)*b^(Floor[(q-1)/n]+1))*Int[x^m*(a+b*x^n)^(p+1)*
      ExpandToSum[n*(p+1)*x^(-m)*Q+Sum[(n*(p+1)+i+1)/a*Coeff[R,x,i]*x^(i-m),{i,0,n-1}],x],x]]] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[n,0] && LtQ[p,-1] && ILtQ[m,0]
```

3: $\int x^m P_q[x^n] \left(a+b x^n\right)^p dx$ when $n \in \mathbb{Z}^+ \land m \in \mathbb{Z} \land \text{GCD}[m+1, n] \neq 1$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z} \land m \in \mathbb{Z}$, let $g = \text{GCD}[m+1, n]$, then $x^m F[x^n] = \frac{1}{g} \text{Subst}\left[x^{\frac{m+1}{g}-1} F\left[x^{\frac{n}{g}}\right], x, x^g\right] \partial_x x^g$

Rule: If $n \in \mathbb{Z}^+ \land m \in \mathbb{Z}$, let $g = \text{GCD}[m+1, n]$, if $g \neq 1$, then

$$\int x^m P_q[x^n] \left(a+b x^n\right)^p dx \rightarrow \frac{1}{g} \text{Subst}\left[\int x^{\frac{m+1}{g}-1} P_q\left[x^{\frac{n}{g}}\right] \left(a+b x^{\frac{n}{g}}\right)^p dx, x, x^g\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{g=GCD[m+1,n]},
  1/g*Subst[Int[x^((m+1)/g-1)*ReplaceAll[Pq,x→x^(1/g)]*(a+b*x^(n/g))^p,x],x,x^g] /;
 g≠1] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x^n] && IGtQ[n,0] && IntegerQ[m]
```

4: $\int \frac{(c\,x)^m P_q[x]}{a+b\,x^n}\,dx$ when $\frac{n}{2}\in\mathbb{Z}^+\wedge q<n$

Derivation: Algebraic expansion

Basis: If $\frac{n}{2}\in\mathbb{Z}\ \wedge\ q<n$, then $P_q[x]=\sum_{i=0}^{n-1}x^i\,P_q[x,i]=\sum_{i=0}^{n/2-1}x^i\left(P_q[x,i]+P_q\left[x,\frac{n}{2}+i\right]x^{n/2}\right)$

Note: The resulting integrands are of the form $\frac{(c\,x)^q\left(r+s\,x^{n/2}\right)}{a+b\,x^n}$ for which there are rules.

Rule: If $\frac{n}{2}\in\mathbb{Z}^+\wedge\ q<n$, then

$$\int\frac{(c\,x)^m P_q[x]}{a+b\,x^n}\,dx\ \to\ \int\sum_{i=0}^{n/2-1}\frac{(c\,x)^{m+i}\left(P_q[x,i]+P_q\left[x,\frac{n}{2}+i\right]x^{n/2}\right)}{c^i\left(a+b\,x^n\right)}\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_/(a_+b_.*x_^n_),x_Symbol] :=
  With[{v=Sum[(c*x)^(m+ii)*(Coeff[Pq,x,ii]+Coeff[Pq,x,n/2+ii]*x^(n/2))/(c^ii*(a+b*x^n)),{ii,0,n/2-1}]},
  Int[v,x] /;
 SumQ[v]] /;
FreeQ[{a,b,c,m},x] && PolyQ[Pq,x] && IGtQ[n/2,0] && Expon[Pq,x]<n
```

5: $\int \frac{P_q[x]}{x\sqrt{a+b\,x^n}}\,dx$ when $n\in\mathbb{Z}^+\land P_q[x,\,0]\neq 0$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+\land P_q[x,\,0]\neq 0$, then

$$\int \frac{P_q[x]}{x\sqrt{a+b\,x^n}}\,dx \longrightarrow P_q[x,\,0]\int \frac{1}{x\sqrt{a+b\,x^n}}\,dx+\int \frac{P_q[x]-P_q[x,\,0]}{x}\,\frac{1}{\sqrt{a+b\,x^n}}\,dx$$

Program code:

```
Int[Pq_/(x_*Sqrt[a_+b_.*x_^n_]),x_Symbol] :=
  Coeff[Pq,x,0]*Int[1/(x*Sqrt[a+b*x^n]),x] +
  Int[ExpandToSum[(Pq-Coeff[Pq,x,0])/x,x]/Sqrt[a+b*x^n],x] /;
FreeQ[{a,b},x] && PolyQ[Pq,x] && IGtQ[n,0] && NeQ[Coeff[Pq,x,0],0]
```

6: $\int (c\,x)^m\, P_q[x]\, \left(a+b\,x^n\right)^p\, dx$ when $\frac{n}{2}\in\mathbb{Z}^+ \land \neg\,\text{PolynomialQ}\left[P_q[x],\, x^{\frac{n}{2}}\right]$

Derivation: Algebraic expansion

Basis: If $n\in\mathbb{Z}^+$, then $P_q[x] == \sum_{j=0}^{n-1} x^j \sum_{k=0}^{(q-j)/n+1} P_q[x,\, j+k\,n]\, x^{k\,n}$

Note: This rule transform integrand into a sum of terms of the form $x^k\, Q_r\left[x^{\frac{n}{2}}\right]\, \left(a+b\,x^n\right)^p$.

Rule: If $\frac{n}{2}\in\mathbb{Z}^+ \land \neg\,\text{PolynomialQ}\left[P_q[x],\, x^{\frac{n}{2}}\right]$, then

$$\int (c\,x)^m\, P_q[x]\, \left(a+b\,x^n\right)^p\, dx \,\longrightarrow\, \int \sum_{j=0}^{\frac{n}{2}-1} \frac{(c\,x)^{m+j}}{c^j} \left(\sum_{k=0}^{\frac{2\,(q-j)}{n}+1} P_q\left[x,\, j+\frac{k\,n}{2}\right] x^{\frac{k\,n}{2}}\right) \left(a+b\,x^n\right)^p\, dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  Module[{q=Expon[Pq,x],j,k},
  Int[Sum[(c*x)^(m+j)/c^j*Sum[Coeff[Pq,x,j+k*n/2]*x^(k*n/2),{k,0,2*(q-j)/n+1}]*(a+b*x^n)^p,{j,0,n/2-1}],x]] /;
FreeQ[{a,b,c,m,p},x] && PolyQ[Pq,x] && IGtQ[n/2,0] && Not[PolyQ[Pq,x^(n/2)]]
```

7: $\int \frac{(c\,x)^m\,P_q[x]}{a+b\,x^n}\,dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+$, then

$$\int \frac{(c\,x)^m\,P_q[x]}{a+b\,x^n}\,dx \longrightarrow \int \text{ExpandIntegrand}\left[\frac{(c\,x)^m\,P_q[x]}{a+b\,x^n},\,x\right]dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_/(a_+b_.*x_^n_),x_Symbol] :=
  Int[ExpandIntegrand[(c*x)^m*Pq/(a+b*x^n),x],x] /;
FreeQ[{a,b,c,m},x] && PolyQ[Pq,x] && IntegerQ[n] && Not[IGtQ[m,0]]
```

8. $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{Z}^+\land q-n\geq -1$

1: $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{Z}^+\land q-n\geq -1\land m<-1\land P_q[x,0]\neq 0$

Derivation: Algebraic expansion and binomial recurrence 3b

Note: This rule increments m *and* decrements the degree of the polynomial in the resulting integrand if $n-1<q$.

Rule: If $n\in\mathbb{Z}^+\land m<-1\land n-1\leq q\land P_q[x,0]\neq 0$, then

$$\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx\,\longrightarrow$$

$$P_q[x,0]\int (c\,x)^m\,\left(a+b\,x^n\right)^p\,dx+\frac{1}{c}\int (c\,x)^{m+1}\,\frac{P_q[x]-P_q[x,0]}{x}\,\left(a+b\,x^n\right)^p\,dx\,\longrightarrow$$

$$\frac{P_q[x,0]\,(c\,x)^{m+1}\,\left(a+b\,x^n\right)^{p+1}}{a\,c\,(m+1)}+\frac{1}{2\,a\,c\,(m+1)}\int (c\,x)^{m+1}\left(2\,a\,(m+1)\,\frac{P_q[x]-P_q[x,0]}{x}-2\,b\,P_q[x,0]\,(m+n\,(p+1)+1)\,x^{n-1}\right)\left(a+b\,x^n\right)^p\,dx$$

Program code:

```
Int[(c_.*x_)^m_*Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{Pq0=Coeff[Pq,x,0]},
    Pq0*(c*x)^(m+1)*(a+b*x^n)^(p+1)/(a*c*(m+1)) +
    1/(2*a*c*(m+1))*Int[(c*x)^(m+1)*ExpandToSum[2*a*(m+1)*(Pq-Pq0)/x-2*b*Pq0*(m+n*(p+1)+1)*x^(n-1),x]*(a+b*x^n)^p,x] /;
 NeQ[Pq0,0]] /;
FreeQ[{a,b,c,p},x] && PolyQ[Pq,x] && IGtQ[n,0] && LtQ[m,-1] && LeQ[n-1,Expon[Pq,x]]
```

2: $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{Z}^+\land\,q-n\geq 0\,\land\,m+q+n\,p+1\neq 0$

Reference: G&R 2.110.5, CRC 88a

Derivation: Algebraic expansion and binomial recurrence 3a

Reference: G&R 2.104

Note: This rule reduces the degree of the polynomial in the resulting integrand.

Rule: If $n\in\mathbb{Z}^+\land\,m+q+n\,p+1\neq 0\,\land\,q-n\geq 0$, then

$$\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx\;\longrightarrow$$

$$\frac{P_q[x,\,q]}{c^q}\int (c\,x)^{m+q}\,\left(a+b\,x^n\right)^p+\int (c\,x)^m\,\left(P_q[x]-P_q[x,\,q]\,x^q\right)\,\left(a+b\,x^n\right)^p\,dx\,dx\;\longrightarrow$$

$$\frac{P_q[x,\,q]\,(c\,x)^{m+q-n+1}\,\left(a+b\,x^n\right)^{p+1}}{b\,c^{q-n+1}\,(m+q+n\,p+1)}+$$

$$\frac{1}{b\,(m+q+n\,p+1)}\int (c\,x)^m\,\left(b\,(m+q+n\,p+1)\,\left(P_q[x]-P_q[x,\,q]\,x^q\right)-a\,P_q[x,\,q]\,(m+q-n+1)\,x^{q-n}\right)\,\left(a+b\,x^n\right)^p\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
    With[{Pqq=Coeff[Pq,x,q]},
    Pqq*(c*x)^(m+q-n+1)*(a+b*x^n)^(p+1)/(b*c^(q-n+1)*(m+q+n*p+1)) +
    1/(b*(m+q+n*p+1))*Int[(c*x)^m*ExpandToSum[b*(m+q+n*p+1)*(Pq-Pqq*x^q)-a*Pqq*(m+q-n+1)*x^(q-n),x]*(a+b*x^n)^p,x]] /;
  NeQ[m+q+n*p+1,0] && q-n≥0 && (IntegerQ[2*p] || IntegerQ[p+(q+1)/(2*n)])] /;
FreeQ[{a,b,c,m,p},x] && PolyQ[Pq,x] && IGtQ[n,0]
```

2. $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{Z}^-$

1. $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{Z}^-\land m\in\mathbb{Q}$

1: $\int x^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{Z}^-\land m\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $F[x] = -\text{Subst}\left[\frac{F\left[x^{-1}\right]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Note: $x^q\,P_q\left[x^{-1}\right]$ is a polynomial in x.

Rule: If $n\in\mathbb{Z}^-\land m\in\mathbb{Z}$, then

$$\int x^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx\;\rightarrow\;-\text{Subst}\left[\int\frac{x^q\,P_q\left[x^{-1}\right]\,\left(a+b\,x^{-n}\right)^p}{x^{m+q+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
  -Subst[Int[ExpandToSum[x^q*ReplaceAll[Pq,x→x^(-1)],x]*(a+b*x^(-n))^p/x^(m+q+2),x],x,1/x]] /;
FreeQ[{a,b,p},x] && PolyQ[Pq,x] && ILtQ[n,0] && IntegerQ[m]
```

2: $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n \in \mathbb{Z}^- \land m \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $g>1$, then $(c\,x)^m\,F[x] = -\frac{g}{c}\,\text{Subst}\left[\frac{F\left[c^{-1}\,x^{-g}\right]}{x^{g\,(m+1)+1}},\,x,\,\frac{1}{(c\,x)^{1/g}}\right]\,\partial_x\,\frac{1}{(c\,x)^{1/g}}$

Note: $x^{g\,q}\,P_q\left[c^{-1}\,x^{-g}\right]$ is a polynomial in x.

Rule: If $n \in \mathbb{Z}^- \land m \in \mathbb{F}$, let $g = \text{Denominator}[m]$, then

$$\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx \,\to\, -\frac{g}{c}\,\text{Subst}\left[\int \frac{x^{g\,q}\,P_q\left[c^{-1}\,x^{-g}\right]\,\left(a+b\,c^{-n}\,x^{-g\,n}\right)^p}{x^{g\,(m+q+1)+1}}\,dx,\,x,\,\frac{1}{(c\,x)^{1/g}}\right]$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{g=Denominator[m],q=Expon[Pq,x]},
  -g/c*Subst[Int[ExpandToSum[x^(g*q)*ReplaceAll[Pq,x→c^(-1)*x^(-g)],x]*
    (a+b*c^(-n)*x^(-g*n))^p/x^(g*(m+q+1)+1),x],x,1/(c*x)^(1/g)]] /;
FreeQ[{a,b,c,p},x] && PolyQ[Pq,x] && ILtQ[n,0] && FractionQ[m]
```

2: $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{Z}^-\wedge m\notin\mathbb{Q}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x\left(\frac{(c\,x)^m}{\left(x^{-1}\right)^m}\right)=0$

Basis: $F[x]=-\text{Subst}\left[\frac{F\left[x^{-1}\right]}{x^2},\,x,\,\frac{1}{x}\right]\partial_x\frac{1}{x}$

Note: $x^q\,P_q\left[x^{-1}\right]$ is a polynomial in x.

Rule: If $n\in\mathbb{Z}^-\wedge m\notin\mathbb{Q}$, then

$$\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx\;\longrightarrow\;(c\,x)^m\,\left(x^{-1}\right)^m\int\frac{P_q[x]\,\left(a+b\,x^n\right)^p}{\left(x^{-1}\right)^m}\,dx$$

$$\longrightarrow\;-(c\,x)^m\,\left(x^{-1}\right)^m\,\text{Subst}\left[\int\frac{x^q\,P_q\left[x^{-1}\right]\,\left(a+b\,x^{-n}\right)^p}{x^{m+q+2}}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(c_.*x_)^m_*Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{q=Expon[Pq,x]},
  -(c*x)^m*(x^(-1))^m*Subst[Int[ExpandToSum[x^q*ReplaceAll[Pq,x→x^(-1)],x]*(a+b*x^(-n))^p/x^(m+q+2),x],x,1/x]] /;
FreeQ[{a,b,c,m,p},x] && PolyQ[Pq,x] && ILtQ[n,0] && Not[RationalQ[m]]
```

8. $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{F}$

1: $\int x^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{F}$

Derivation: Integration by substitution

Basis: If $g\in\mathbb{Z}^+$, then $x^m\,P_q[x]\,F[x^n] == g\,\text{Subst}\left[x^{g\,(m+1)-1}\,P_q[x^g]\,F[x^{g\,n}],\,x,\,x^{1/g}\right]\partial_x x^{1/g}$

Rule: If $n\in\mathbb{F}$, let $g=\text{Denominator}[n]$, then

$$\int x^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx\;\longrightarrow\;g\,\text{Subst}\left[\int x^{g\,(m+1)-1}\,P_q\left[x^g\right]\,\left(a+b\,x^{g\,n}\right)^p\,dx,\,x,\,x^{1/g}\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{g=Denominator[n]},
  g*Subst[Int[x^(g*(m+1)-1)*ReplaceAll[Pq,x→x^g]*(a+b*x^(g*n))^p,x],x,x^(1/g)]] /;
FreeQ[{a,b,m,p},x] && PolyQ[Pq,x] && FractionQ[n]
```

2: $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{F}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{(c\,x)^m}{x^m}==0$

Basis: $\frac{(c\,x)^m}{x^m}==\frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule: If $n\in\mathbb{F}$, then

$$\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx\;\longrightarrow\;\frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$$

Program code:

```
Int[(c_*x_)^m_*Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*Pq*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,m,p},x] && PolyQ[Pq,x] && FractionQ[n]
```

9. $\int (c\,x)^m\,P_q\left[x^n\right]\,\left(a+b\,x^n\right)^p\,dx$ when $\frac{n}{m+1}\in\mathbb{Z}$

1: $\int x^m\,P_q\left[x^n\right]\,\left(a+b\,x^n\right)^p\,dx$ when $\frac{n}{m+1}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{n}{m+1}\in\mathbb{Z}$, then $x^m\,F\left[x^n\right]=\frac{1}{m+1}\,\text{Subst}\left[F\left[x^{\frac{n}{m+1}}\right],\,x,\,x^{m+1}\right]\,\partial_x\,x^{m+1}$

Rule: If $\frac{n}{m+1}\in\mathbb{Z}$

$$\int x^m\,P_q\left[x^n\right]\,\left(a+b\,x^n\right)^p\,dx\;\longrightarrow\;\frac{1}{m+1}\,\text{Subst}\left[\int P_q\left[x^{\frac{n}{m+1}}\right]\,\left(a+b\,x^{\frac{n}{m+1}}\right)^p\,dx,\,x,\,x^{m+1}\right]$$

Program code:

```
Int[x_^m_.*Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  1/(m+1)*Subst[Int[ReplaceAll[SubstFor[x^n,Pq,x],x→x^Simplify[n/(m+1)]]*(a+b*x^Simplify[n/(m+1)])^p,x],x,x^(m+1)] /;
FreeQ[{a,b,m,n,p},x] && PolyQ[Pq,x^n] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

2: $\int (c\,x)^m\,P_q[x^n]\,(a+b\,x^n)^p\,dx$ when $\frac{n}{m+1}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m} == 0$

Basis: $\frac{(c\,x)^m}{x^m} == \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}$

Rule: If $\frac{n}{m+1}\in\mathbb{Z}$, then

$$\int (c\,x)^m\,P_q[x^n]\,(a+b\,x^n)^p\,dx \longrightarrow \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,P_q[x^n]\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[(c_*x_)^m_*Pq_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*Pq*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,m,n,p},x] && PolyQ[Pq,x^n] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

10: $\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx$

Derivation: Algebraic expansion

Rule:

$$\int (c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p\,dx\;\longrightarrow\;\int \text{ExpandIntegrand}\left[(c\,x)^m\,P_q[x]\,\left(a+b\,x^n\right)^p,\,x\right]dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a_+b_.*x_^n_)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(c*x)^m*Pq*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,m,n,p},x] && (PolyQ[Pq,x] || PolyQ[Pq,x^n]) && Not[IGtQ[m,0]]
```

S: $\int u^m\,P_q[v^n]\,\left(a+b\,v^n\right)^p\,dx$ when $v == f+g\,x \land u == h\,v$

Derivation: Integration by substitution and piecewise constant extraction

Basis: If $u == h\,v$, then $\partial_x \frac{u^m}{v^m} == 0$

Rule: If $v == f+g\,x \land u == h\,v$, then

$$\int u^m\,P_q[v^n]\,\left(a+b\,v^n\right)^p\,dx\;\longrightarrow\;\frac{u^m}{g\,v^m}\,\text{Subst}\left[\int x^m\,P_q[x^n]\,\left(a+b\,x^n\right)^p\,dx,\,x,\,v\right]$$

Program code:

```
Int[u_^m_.*Pq_*(a_+b_.*v_^n_.)^p_,x_Symbol] :=
  u^m/(Coeff[v,x,1]*v^m)*Subst[Int[x^m*SubstFor[v,Pq,x]*(a+b*x^n)^p,x],x,v] /;
FreeQ[{a,b,m,n,p},x] && LinearPairQ[u,v,x] && PolyQ[Pq,v^n]
```

Rules for integrands of the form $(h\,x)^m\,P_q[x]\,(a+b\,x^n)^p\,(c+d\,x^n)^q$

1. $\int (c\,x)^m\,P_q[x]\,\left(a_1+b_1\,x^n\right)^p\,\left(a_2+b_2\,x^n\right)^p\,dx$ when $a_2\,b_1+a_1\,b_2=0$

1: $\int (c\,x)^m\,P_q[x]\,\left(a_1+b_1\,x^n\right)^p\,\left(a_2+b_2\,x^n\right)^p\,dx$ when $a_2\,b_1+a_1\,b_2=0\,\wedge\,(p\in\mathbb{Z}\,\vee\,a_1>0\,\wedge\,a_2>0)$

Derivation: Algebraic simplification

Basis: If $a_2\,b_1+a_1\,b_2=0\,\wedge\,(p\in\mathbb{Z}\,\vee\,a_1>0\,\wedge\,a_2>0)$, then $\left(a_1+b_1\,x^n\right)^p\,\left(a_2+b_2\,x^n\right)^p=\left(a_1\,a_2+b_1\,b_2\,x^{2\,n}\right)^p$

Rule: If $a_2\,b_1+a_1\,b_2=0\,\wedge\,(p\in\mathbb{Z}\,\vee\,a_1>0\,\wedge\,a_2>0)$, then

$$\int (c\,x)^m\,P_q[x]\,\left(a_1+b_1\,x^n\right)^p\,\left(a_2+b_2\,x^n\right)^p\,dx\;\longrightarrow\;\int (c\,x)^m\,P_q[x]\,\left(a_1\,a_2+b_1\,b_2\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a1_+b1_.*x_^n_.)^p_.*(a2_+b2_.*x_^n_.)^p_.,x_Symbol] :=
  Int[(c*x)^m*Pq*(a1*a2+b1*b2*x^(2*n))^p,x] /;
FreeQ[{a1,b1,a2,b2,c,m,n,p},x] && PolyQ[Pq,x] && EqQ[a2*b1+a1*b2,0] && (IntegerQ[p] || GtQ[a1,0] && GtQ[a2,0])
```

2: $\int (c\,x)^m\,P_q[x]\,\left(a_1+b_1\,x^n\right)^p\,\left(a_2+b_2\,x^n\right)^p\,dx$ when $a_2\,b_1+a_1\,b_2=0$

Derivation: Piecewise constant extraction

Basis: If $a_2\,b_1+a_1\,b_2=0$, then $\partial_x\,\frac{\left(a_1+b_1\,x^n\right)^p\,\left(a_2+b_2\,x^n\right)^p}{\left(a_1\,a_2+b_1\,b_2\,x^{2\,n}\right)^p}=0$

Rule: If $a_2\,b_1+a_1\,b_2=0$, then

$$\int (c\,x)^m\,P_q[x]\,\left(a_1+b_1\,x^n\right)^p\,\left(a_2+b_2\,x^n\right)^p\,dx\;\to\;\frac{\left(a_1+b_1\,x^n\right)^{\text{FracPart}[p]}\,\left(a_2+b_2\,x^n\right)^{\text{FracPart}[p]}}{\left(a_1\,a_2+b_1\,b_2\,x^{2\,n}\right)^{\text{FracPart}[p]}}\int (c\,x)^m\,P_q[x]\,\left(a_1\,a_2+b_1\,b_2\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*Pq_*(a1_+b1_.*x_^n_.)^p_.*(a2_+b2_.*x_^n_.)^p_.,x_Symbol] :=
  (a1+b1*x^n)^FracPart[p]*(a2+b2*x^n)^FracPart[p]/(a1*a2+b1*b2*x^(2*n))^FracPart[p]*
    Int[(c*x)^m*Pq*(a1*a2+b1*b2*x^(2*n))^p,x] /;
FreeQ[{a1,b1,a2,b2,c,m,n,p},x] && PolyQ[Pq,x] && EqQ[a2*b1+a1*b2,0] && Not[EqQ[n,1] && LinearQ[Pq,x]]
```

2: $\int (h\,x)^m\,\left(e+f\,x^n+g\,x^{2\,n}\right)\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^p\,dx$ when $a\,c\,f\,(m+1)==e\,\left(b\,c+a\,d\right)\,(m+n\,(p+1)+1)\,\land\,a\,c\,g\,(m+1)==b\,d\,e\,(m+2\,n\,(p+1)+1)\,\land\,m\neq-1$

Rule: If

$a\,c\,f\,(m+1)==e\,\left(b\,c+a\,d\right)\,(m+n\,(p+1)+1)\,\land\,a\,c\,g\,(m+1)==b\,d\,e\,(m+2\,n\,(p+1)+1)\,\land\,m\neq-1$, then

$$\int (h\,x)^m\,\left(e+f\,x^n+g\,x^{2\,n}\right)\,\left(a+b\,x^n\right)^p\,\left(c+d\,x^n\right)^p\,dx\;\to\;\frac{e\,(h\,x)^{m+1}\,\left(a+b\,x^n\right)^{p+1}\,\left(c+d\,x^n\right)^{p+1}}{a\,c\,h\,(m+1)}$$

Program code:

```
Int[(h_.*x_)^m_.*(e_+f_.*x_^n_.+g_.*x_^n2_.)*(a_+b_.*x_^n_.)^p_.*(c_+d_.*x_^n_.)^p_.,x_Symbol] :=
  e*(h*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(p+1)/(a*c*h*(m+1)) /;
FreeQ[{a,b,c,d,e,f,g,h,m,n,p},x] && EqQ[n2,2*n] && EqQ[a*c*f*(m+1)-e*(b*c+a*d)*(m+n*(p+1)+1),0] &&
  EqQ[a*c*g*(m+1)-b*d*e*(m+2*n*(p+1)+1),0] && NeQ[m,-1]
```

```
Int[(h_.*x_)^m_.*(e_+g_.*x_^n2_.)*(a_+b_.*x_^n_.)^p_.*(c_+d_.*x_^n_.)^p_.,x_Symbol] :=
  e*(h*x)^(m+1)*(a+b*x^n)^(p+1)*(c+d*x^n)^(p+1)/(a*c*h*(m+1)) /;
FreeQ[{a,b,c,d,e,g,h,m,n,p},x] && EqQ[n2,2*n] && EqQ[m+n*(p+1)+1,0] && EqQ[a*c*g*(m+1)-b*d*e*(m+2*n*(p+1)+1),0] &&
  NeQ[m,-1]
```