Rules for integrands of the form $(d + e x^2)^q (a + b x^2 + c x^4)^p$

0. $\int (d + e x^2)^q (b x^2 + c x^4)^p \, dx$ when $p \notin \mathbb{Z}$

   1. $\int (d + e x^2) (b x^2 + c x^4)^p \, dx$ when $p \notin \mathbb{Z}$

      1: $\int \frac{d + e x^2}{(b x^2 + c x^4)^{3/4}} \, dx$

Derivation: Trinomial recurrence 2a with $a = 0, m = 0$ and $n (2 p + 1) + 1 == 0$ composed with trinomial recurrene 5 with $a = 0$

Rule 1.2.2.3.0.1.1:

$$\int \frac{d + e x^2}{(b x^2 + c x^4)^{3/4}} \, dx \longrightarrow -\frac{2 (c d - b e) (b x^2 + c x^4)^{1/4}}{b c x} + \frac{e}{c} \int \frac{(b x^2 + c x^4)^{1/4}}{x^2} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)/(b_.*x_^2+c_.*x_^4)^(3/4),x_Symbol] :=
  -2*(c*d-b*e)*(b*x^2+c*x^4)^(1/4)/(b*c*x) + e/c*Int[(b*x^2+c*x^4)^(1/4)/x^2,x] /;
FreeQ[{b,c,d,e},x]
```

2: $\int (d+e x^2) (b x^2+c x^4)^p \, dx$ when $p \notin \mathbb{Z} \land p \neq -\frac{3}{4} \land b e (2p+1) - c d (4p+3) == 0$

Derivation: Trinomial recurrence 3a with $a = 0$ with $b e (n p+1) - c d (n (2p+1)+1) == 0$

Rule 1.2.2.3.0.1.2: If $p \notin \mathbb{Z} \land p \neq -\frac{3}{4} \land b e (2p+1) - c d (4p+3) == 0$, then

$$\int (d+e x^2) (b x^2+c x^4)^p \, dx \to \frac{e (b x^2+c x^4)^{p+1}}{c (4p+3) x}$$

Program code:

```
Int[(d_+e_.*x_^2)*(b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  e*(b*x^2+c*x^4)^(p+1)/(c*(4*p+3)*x) /;
FreeQ[{b,c,d,e,p},x] && Not[IntegerQ[p]] && NeQ[4*p+3,0] && EqQ[b*e*(2*p+1)-c*d*(4*p+3),0]
```

3: $\int (d+e x^2) (b x^2+c x^4)^p \, dx$ when $p \notin \mathbb{Z} \land p \neq -\frac{3}{4} \land b e (2p+1) - c d (4p+3) \neq 0$

Derivation: Trinomial recurrence 3a with $a = 0$

Rule 1.2.2.3.0.1.3: If $p \notin \mathbb{Z} \land p \neq -\frac{3}{4} \land b e (2p+1) - c d (4p+3) \neq 0$, then

$$\int (d+e x^2) (b x^2+c x^4)^p \, dx \to \frac{e (b x^2+c x^4)^{p+1}}{c (4p+3) x} - \frac{b e (2p+1) - c d (4p+3)}{c (4p+3)} \int (b x^2+c x^4)^p \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)*(b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  e*(b*x^2+c*x^4)^(p+1)/(c*(4*p+3)*x) - ((b*e*(2*p+1)-c*d*(4*p+3))/(c*(4*p+3)))*Int[(b*x^2+c*x^4)^p,x] /;
FreeQ[{b,c,d,e,p},x] && Not[IntegerQ[p]] && NeQ[4*p+3,0] && NeQ[b*e*(2*p+1)-c*d*(4*p+3),0]
```

2: $\int (d+e\,x^2)^q\,(b\,x^2+c\,x^4)^p\,dx$ when $p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(b\,x^2+c\,x^4)^p}{x^{2\,p}\,(b+c\,x^2)^p} == 0$

Basis: $\frac{(b\,x^2+c\,x^4)^{\text{FracPart}[p]}}{x^{2\,\text{FracPart}[p]}\,(b+c\,x^2)^{\text{FracPart}[p]}} == \frac{(b\,x^2+c\,x^4)^{\text{FracPart}[p]}}{x^{2\,\text{FracPart}[p]}\,(b+c\,x^2)^{\text{FracPart}[p]}}$

Rule 1.2.2.3.0.2: If $p\notin\mathbb{Z}$, then

$$\int (d+e\,x^2)^q\,(b\,x^2+c\,x^4)^p\,dx \rightarrow \frac{(b\,x^2+c\,x^4)^{\text{FracPart}[p]}}{x^{2\,\text{FracPart}[p]}\,(b+c\,x^2)^{\text{FracPart}[p]}}\int x^{2\,p}\,(d+e\,x^2)^q\,(b+c\,x^2)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_.*(b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (b*x^2+c*x^4)^FracPart[p]/(x^(2*FracPart[p])*(b+c*x^2)^FracPart[p])*Int[x^(2*p)*(d+e*x^2)^q*(b+c*x^2)^p,x] /;
FreeQ[{b,c,d,e,p,q},x] && Not[IntegerQ[p]]
```

1. $\int (d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c==0$

x: $\int (d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c==0\,\wedge\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2-4\,a\,c==0$, then $a+b\,z+c\,z^2==\frac{1}{c}\left(\frac{b}{2}+c\,z\right)^2$

Rule 1.2.2.2.1.x: If $b^2-4\,a\,c==0\,\wedge\,p\in\mathbb{Z}$, then

$$\int (d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx\,\longrightarrow\,\frac{1}{c^p}\int (d+e\,x^2)^q\left(\frac{b}{2}+c\,x^2\right)^{2\,p}dx$$

Program code:

```
(* Int[(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  1/c^p*Int[(d+e*x^2)^q*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,e,p,q},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p] *)
```

2. $\int (d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}$

1: $\int (d+e\,x^2)^q\,(a+b\,x^2+c\,x^4)^p\,dx$ when $b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,2\,c\,d-b\,e==0$ **Necessary ??**

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c==0\,\wedge\,2\,c\,d-b\,e==0$, then $\partial_x\frac{(a+b\,x^2+c\,x^4)^p}{(d+e\,x^2)^{2\,p}}==0$

Note: If $b^2-4\,a\,c==0\,\wedge\,2\,c\,d-b\,e==0$, then $a+b\,z+c\,z^2==\frac{c}{e^2}\,(d+e\,z)^2$

Rule 1.2.2.3.1.2.1: If $b^2-4\,a\,c==0\,\wedge\,p\notin\mathbb{Z}\,\wedge\,2\,c\,d-b\,e==0$, then

$$\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \rightarrow \frac{(a+b x^2+c x^4)^p}{(d+e x^2)^{2p}} \int (d+e x^2)^{q+2p} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (a+b*x^2+c*x^4)^p/(d+e*x^2)^(2*p)*Int[(d+e*x^2)^(q+2*p),x] /;
FreeQ[{a,b,c,d,e,p,q},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]] && EqQ[2*c*d-b*e,0]
```

2: $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c == 0 \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4 a c == 0$, then $\partial_x \frac{(a+b x^2+c x^4)^p}{\left(\frac{b}{2}+c x^2\right)^{2p}} == 0$

Note: If $b^2-4 a c == 0$, then $a+b z+c z^2 == \frac{1}{c}\left(\frac{b}{2}+c z\right)^2$

Rule 1.2.2.3.1.2.2: If $b^2-4 a c == 0 \land p \notin \mathbb{Z}$, then

$$\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \rightarrow \frac{(a+b x^2+c x^4)^{\text{FracPart}[p]}}{c^{\text{IntPart}[p]} \left(\frac{b}{2}+c x^2\right)^{2\,\text{FracPart}[p]}} \int (d+e x^2)^q \left(\frac{b}{2}+c x^2\right)^{2p} dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (a+b*x^2+c*x^4)^FracPart[p]/(c^IntPart[p]*(b/2+c*x^2)^(2*FracPart[p]))*Int[(d+e*x^2)^q*(b/2+c*x^2)^(2*p),x] /;
FreeQ[{a,b,c,d,e,p,q},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2. $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0$

1: $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $c d^2-b d e+a e^2 == 0$, then $a+b z+c z^2 == (d+e z) \left(\frac{a}{d}+\frac{c z}{e}\right)$

Rule 1.2.2.3.2.1: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \in \mathbb{Z}$, then

$$\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \to \int (d+e x^2)^{p+q} \left(\frac{a}{d}+\frac{c x^2}{e}\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Int[(d+e*x^2)^(p+q)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,q},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[p]
```

```
Int[(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Int[(d+e*x^2)^(p+q)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,c,d,e,q},x] && EqQ[c*d^2+a*e^2,0] && IntegerQ[p]
```

2: $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 == 0 \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $c d^2-b d e+a e^2 == 0$, then $\partial_x \frac{(a+b x^2+c x^4)^p}{(d+e x^2)^p \left(\frac{a}{d}+\frac{c x^2}{e}\right)^p} == 0$

Basis: If $c d^2 - b d e + a e^2 == 0$, then $\frac{(a+b x^2+c x^4)^p}{(d+e x^2)^p \left(\frac{a}{d}+\frac{c x^2}{e}\right)^p} == \frac{(a+b x^2+c x^4)^{\text{FracPart}[p]}}{(d+e x^2)^{\text{FracPart}[p]} \left(\frac{a}{d}+\frac{c x^2}{e}\right)^{\text{FracPart}[p]}}$

Rule 1.2.2.3.2.2: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 == 0 \land p \notin \mathbb{Z}$, then

$$\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \to \frac{(a+b x^2+c x^4)^{\text{FracPart}[p]}}{(d+e x^2)^{\text{FracPart}[p]} \left(\frac{a}{d}+\frac{c x^2}{e}\right)^{\text{FracPart}[p]}} \int (d+e x^2)^{p+q} \left(\frac{a}{d}+\frac{c x^2}{e}\right)^p dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  (a+b*x^2+c*x^4)^FracPart[p]/((d+e*x^2)^FracPart[p]*(a/d+c*x^2/e)^FracPart[p])*Int[(d+e*x^2)^(p+q)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,b,c,d,e,p,q},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[p]]
```

```
Int[(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_,x_Symbol] :=
  (a+c*x^4)^FracPart[p]/((d+e*x^2)^FracPart[p]*(a/d+c*x^2/e)^FracPart[p])*Int[(d+e*x^2)^(p+q)*(a/d+c/e*x^2)^p,x] /;
FreeQ[{a,c,d,e,p,q},x] && EqQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]]
```

3. $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p \in \mathbb{Z}^+$

1: $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p \in \mathbb{Z}^+ \land q+2 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.2.2.3.3.1: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p \in \mathbb{Z}^+ \land q+2 \in \mathbb{Z}^+$, then

$$\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \to \int \text{ExpandIntegrand}\left[(d+e x^2)^q (a+b x^2+c x^4)^p, x\right] dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^q*(a+b*x^2+c*x^4)^p,x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[p,0] && IGtQ[q,-2]
```

```
Int[(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^q*(a+c*x^4)^p,x],x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && IGtQ[p,0] && IGtQ[q,-2]
```

2. $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge p \in \mathbb{Z}^+ \wedge q < -1$

1: $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge p \in \mathbb{Z}^+ \wedge q+\frac{1}{2} \in \mathbb{Z}^- \wedge 4 p+2 q+1 < 0$

Derivation: Algebraic expansion and binomial recurrence 3b

Basis: $\int (d+e x^2)^q \, dx = \frac{x (d+e x^2)^{q+1}}{d} - \frac{e (2 q+3)}{d} \int x^2 (d+e x^2)^q \, dx$

Note: Interestingly this rule eleminates the constant term of $(a+b x^2+c x^4)^p$ rather than the highest degree term.

Rule 1.2.2.3.3.2.1: If $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge p \in \mathbb{Z}^+ \wedge q+\frac{1}{2} \in \mathbb{Z}^- \wedge 4 p+2 q+1 < 0$, then

$$\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \longrightarrow$$

$$a^p \int (d+e x^2)^q \, dx + \int x^2 (d+e x^2)^q \, \text{PolynomialQuotient}\left[(a+b x^2+c x^4)^p - a^p, x^2, x\right] dx \longrightarrow$$

$$\frac{a^p x (d+e x^2)^{q+1}}{d} + \frac{1}{d} \int x^2 (d+e x^2)^q \left(d \, \text{PolynomialQuotient}\left[(a+b x^2+c x^4)^p - a^p, x^2, x\right] - e a^p (2 q+3)\right) dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  a^p*x*(d+e*x^2)^(q+1)/d +
  1/d*Int[x^2*(d+e*x^2)^q*(d*PolynomialQuotient[(a+b*x^2+c*x^4)^p-a^p,x^2,x]-e*a^p*(2*q+3)),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[p,0] && ILtQ[q+1/2,0] && LtQ[4*p+2*q+1,0]
```

```
Int[(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  a^p*x*(d+e*x^2)^(q+1)/d +
  1/d*Int[x^2*(d+e*x^2)^q*(d*PolynomialQuotient[(a+c*x^4)^p-a^p,x^2,x]-e*a^p*(2*q+3)),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && IGtQ[p,0] && ILtQ[q+1/2,0] && LtQ[4*p+2*q+1,0]
```

2: $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p \in \mathbb{Z}^+ \land q < -1$

Derivation: Algebraic expansion and quadratic recurrence 2a

Rule 1.2.2.3.3.2.2: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p \in \mathbb{Z}^+ \land q < -1$,
let $Q[x] \to \text{PolynomialQuotient}\left[(a+b x^2+c x^4)^p, d+e x^2, x\right]$ and
$R \to \text{PolynomialRemainder}\left[(a+b x^2+c x^4)^p, d+e x^2, x\right]$, then

$$\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \longrightarrow$$

$$R \int (d+e x^2)^q \, dx + \int Q[x] (d+e x^2)^{q+1} \, dx \longrightarrow$$

$$-\frac{R x (d+e x^2)^{q+1}}{2 d (q+1)} + \frac{1}{2 d (q+1)} \int (d+e x^2)^{q+1} \left(2 d (q+1) Q[x] + R (2 q+3)\right) dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  With[{Qx=PolynomialQuotient[(a+b*x^2+c*x^4)^p,d+e*x^2,x],
        R=Coeff[PolynomialRemainder[(a+b*x^2+c*x^4)^p,d+e*x^2,x],x,0]},
  -R*x*(d+e*x^2)^(q+1)/(2*d*(q+1)) +
  1/(2*d*(q+1))*Int[(d+e*x^2)^(q+1)*ExpandToSum[2*d*(q+1)*Qx+R*(2*q+3),x],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[p,0] && LtQ[q,-1]
```

```
Int[(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  With[{Qx=PolynomialQuotient[(a+c*x^4)^p,d+e*x^2,x],
        R=Coeff[PolynomialRemainder[(a+c*x^4)^p,d+e*x^2,x],x,0]},
  -R*x*(d+e*x^2)^(q+1)/(2*d*(q+1)) +
  1/(2*d*(q+1))*Int[(d+e*x^2)^(q+1)*ExpandToSum[2*d*(q+1)*Qx+R*(2*q+3),x],x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && IGtQ[p,0] && LtQ[q,-1]
```

3: $\int \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,p\in\mathbb{Z}^+\land\,q\nless -1$

Reference: G&R 2.110.5, G&R 2.104, G&R 2.160.3, CRC 88a

Derivation: Algebraic expansion and binomial recurrence 3a

Note: If $p\in\mathbb{Z}^+\land\,q\nless -1$, then $4\,p+2\,q+1\neq 0$.

Rule 1.2.2.3.3.3: If $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-b\,d\,e+a\,e^2\neq 0\,\land\,p\in\mathbb{Z}^+\land\,q\nless -1$, then

$$\int \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p\,dx\,\longrightarrow$$

$$c^p\int x^{4\,p}\left(d+e\,x^2\right)^q dx+\int \left(d+e\,x^2\right)^q \left(\left(a+b\,x^2+c\,x^4\right)^p-c^p\,x^{4\,p}\right)dx\,\longrightarrow$$

$$\frac{c^p\,x^{4\,p-1}\left(d+e\,x^2\right)^{q+1}}{e\,(4\,p+2\,q+1)}+\frac{1}{e\,(4\,p+2\,q+1)}\int \left(d+e\,x^2\right)^q \left(e\,(4\,p+2\,q+1)\,\left(a+b\,x^2+c\,x^4\right)^p-d\,c^p\,(4\,p-1)\,x^{4\,p-2}-e\,c^p\,(4\,p+2\,q+1)\,x^{4\,p}\right)dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  c^p*x^(4*p-1)*(d+e*x^2)^(q+1)/(e*(4*p+2*q+1)) +
  1/(e*(4*p+2*q+1))*Int[(d+e*x^2)^q*ExpandToSum[e*(4*p+2*q+1)*(a+b*x^2+c*x^4)^p-d*c^p*(4*p-1)*x^(4*p-2)-e*c^p*(4*p+2*q+1)*x^(4*p
FreeQ[{a,b,c,d,e,q},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[p,0] && Not[LtQ[q,-1]]
```

```
Int[(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  c^p*x^(4*p-1)*(d+e*x^2)^(q+1)/(e*(4*p+2*q+1)) +
  1/(e*(4*p+2*q+1))*Int[(d+e*x^2)^q*ExpandToSum[e*(4*p+2*q+1)*(a+c*x^4)^p-d*c^p*(4*p-1)*x^(4*p-2)-e*c^p*(4*p+2*q+1)*x^(4*p),x],x
FreeQ[{a,c,d,e,q},x] && NeQ[c*d^2+a*e^2,0] && IGtQ[p,0] && Not[LtQ[q,-1]]
```

4. $\int \frac{(d+e x^2)^q}{a+b x^2+c x^4} dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$

1. $\int \frac{(d+e x^2)^q}{a+b x^2+c x^4} dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q \in \mathbb{Z}$

1. $\int \frac{d+e x^2}{a+b x^2+c x^4} dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$

1. $\int \frac{d+e x^2}{a+b x^2+c x^4} dx$ when $b^2-4 a c \neq 0 \land c d^2-a e^2 == 0$

1: $\int \frac{d+e x^2}{a+b x^2+c x^4} dx$ when $b^2-4 a c \neq 0 \land c d^2-a e^2 == 0 \land \frac{2 d}{e}-\frac{b}{c}>0$

Derivation: Algebraic expansion

Basis: If $c d^2-a e^2 == 0$ and $q \to \sqrt{\frac{2 d}{e}-\frac{b}{c}}$, then $\frac{d+e z^2}{a+b z^2+c z^4} == \frac{e^2}{2 c (d+e q z+e z^2)}+\frac{e^2}{2 c (d-e q z+e z^2)}$

Rule 1.2.2.3.4.1.1.1.1: If $b^2-4 a c \neq 0 \land c d^2-a e^2 == 0 \land \frac{2 d}{e}-\frac{b}{c}>0$, let $q \to \sqrt{\frac{2 d}{e}-\frac{b}{c}}$, then

$$\int \frac{d+e x^2}{a+b x^2+c x^4} dx \to \frac{e}{2 c} \int \frac{1}{\frac{d}{e}+q x+x^2} dx+\frac{e}{2 c} \int \frac{1}{\frac{d}{e}-q x+x^2} dx$$

Program code:

```
Int[(d_+e_.*x_^2)/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[2*d/e-b/c,2]},
  e/(2*c)*Int[1/Simp[d/e+q*x+x^2,x],x] + e/(2*c)*Int[1/Simp[d/e-q*x+x^2,x],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-a*e^2,0] && (GtQ[2*d/e-b/c,0] || Not[LtQ[2*d/e-b/c,0]] && EqQ[d-e*Rt[a/c,2
```

```
Int[(d_+e_.*x_^2)/(a_+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[2*d/e,2]},
  e/(2*c)*Int[1/Simp[d/e+q*x+x^2,x],x] + e/(2*c)*Int[1/Simp[d/e-q*x+x^2,x],x]] /;
FreeQ[{a,c,d,e},x] && EqQ[c*d^2-a*e^2,0] && PosQ[d*e]
```

2: $\int \frac{d+e x^2}{a+b x^2+c x^4} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-a e^2 == 0 \land b^2-4 a c > 0$

Derivation: Algebraic expansion

- Basis: Let $q \to \sqrt{b^2-4 a c}$, then $\frac{d+e z}{a+b z+c z^2} == \left(\frac{e}{2}+\frac{2 c d-b e}{2 q}\right) \frac{1}{\frac{b}{2}-\frac{q}{2}+c z}+\left(\frac{e}{2}-\frac{2 c d-b e}{2 q}\right) \frac{1}{\frac{b}{2}+\frac{q}{2}+c z}$

Rule 1.2.2.3.4.1.1.1.2: If $b^2-4 a c \neq 0 \land c d^2-a e^2 == 0 \land b^2-4 a c > 0$, let $q \to \sqrt{b^2-4 a c}$, then

$$\int \frac{d+e x^2}{a+b x^2+c x^4} \, dx \longrightarrow \left(\frac{e}{2}+\frac{2 c d-b e}{2 q}\right) \int \frac{1}{\frac{b}{2}-\frac{q}{2}+c x^2} \, dx+\left(\frac{e}{2}-\frac{2 c d-b e}{2 q}\right) \int \frac{1}{\frac{b}{2}+\frac{q}{2}+c x^2} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (e/2+(2*c*d-b*e)/(2*q))*Int[1/(b/2-q/2+c*x^2),x] + (e/2-(2*c*d-b*e)/(2*q))*Int[1/(b/2+q/2+c*x^2),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-a*e^2,0] && GtQ[b^2-4*a*c,0]
```

3: $\int \frac{d+e x^2}{a+b x^2+c x^4} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-a e^2 == 0 \land b^2-4 a c \ngtr 0$

Derivation: Algebraic expansion

Basis: If $c d^2-a e^2 == 0$ and $q \to \sqrt{-\frac{2 d}{e}-\frac{b}{c}}$, then $\frac{d+e z^2}{a+b z^2+c z^4} == \frac{e (q-2 z)}{2 c q \left(\frac{d}{e}+q z-z^2\right)}+\frac{e (q+2 z)}{2 c q \left(\frac{d}{e}-q z-z^2\right)}$

Rule 1.2.2.3.4.1.1.1.3: If $b^2-4 a c \neq 0 \land c d^2-a e^2 == 0 \land b^2-4 a c \ngtr 0$, let $q \to \sqrt{-\frac{2 d}{e}-\frac{b}{c}}$, then

$$\int \frac{d+e x^2}{a+b x^2+c x^4} \, dx \to \frac{e}{2 c q} \int \frac{q-2 x}{\frac{d}{e}+q x-x^2} \, dx+\frac{e}{2 c q} \int \frac{q+2 x}{\frac{d}{e}-q x-x^2} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[-2*d/e-b/c,2]},
  e/(2*c*q)*Int[(q-2*x)/Simp[d/e+q*x-x^2,x],x] + e/(2*c*q)*Int[(q+2*x)/Simp[d/e-q*x-x^2,x],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && EqQ[c*d^2-a*e^2,0] && Not[GtQ[b^2-4*a*c,0]]
```

```
Int[(d_+e_.*x_^2)/(a_+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[-2*d/e,2]},
  e/(2*c*q)*Int[(q-2*x)/Simp[d/e+q*x-x^2,x],x] + e/(2*c*q)*Int[(q+2*x)/Simp[d/e-q*x-x^2,x],x]] /;
FreeQ[{a,c,d,e},x] && EqQ[c*d^2-a*e^2,0] && NegQ[d*e]
```

2. $\int \frac{d+e x^2}{a+b x^2+c x^4}\,dx$ when $b^2-4ac\neq 0 \land c d^2-a e^2\neq 0$

1: $\int \frac{d+e x^2}{a+b x^2+c x^4}\,dx$ when $b^2-4ac\neq 0 \land c d^2-a e^2\neq 0 \land b^2-4ac>0$

Derivation: Algebraic expansion

- Basis: Let $q\to\sqrt{b^2-4ac}$, then $\frac{d+e z}{a+b z+c z^2} == \left(\frac{e}{2}+\frac{2cd-be}{2q}\right)\frac{1}{\frac{b}{2}-\frac{q}{2}+c z}+\left(\frac{e}{2}-\frac{2cd-be}{2q}\right)\frac{1}{\frac{b}{2}+\frac{q}{2}+c z}$

- Rule 1.2.2.3.4.1.1.2.1: If $b^2-4ac\neq 0 \land c d^2-a e^2\neq 0 \land b^2-4ac>0$, let $q\to\sqrt{b^2-4ac}$, then

$$\int \frac{d+e x^2}{a+b x^2+c x^4}\,dx \to \left(\frac{e}{2}+\frac{2cd-be}{2q}\right)\int \frac{1}{\frac{b}{2}-\frac{q}{2}+c x^2}\,dx+\left(\frac{e}{2}-\frac{2cd-be}{2q}\right)\int \frac{1}{\frac{b}{2}+\frac{q}{2}+c x^2}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (e/2+(2*c*d-b*e)/(2*q))*Int[1/(b/2-q/2+c*x^2),x] + (e/2-(2*c*d-b*e)/(2*q))*Int[1/(b/2+q/2+c*x^2),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-a*e^2,0] && PosQ[b^2-4*a*c]
```

```
Int[(d_+e_.*x_^2)/(a_+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  (e/2+c*d/(2*q))*Int[1/(-q+c*x^2),x] + (e/2-c*d/(2*q))*Int[1/(q+c*x^2),x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2-a*e^2,0] && PosQ[-a*c]
```

2. $\int \frac{d+e\,x^2}{a+b\,x^2+c\,x^4}\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,c\,d^2-a\,e^2\neq 0$

1: $\int \frac{d+e\,x^2}{a+c\,x^4}\,dx$ when $c\,d^2+a\,e^2\neq 0\,\land\,c\,d^2-a\,e^2\neq 0\,\land\,-a\,c\not>0$

Derivation: Algebraic expansion

Basis: Let $q\to\sqrt{a\,c}$, then $\frac{d+e\,z}{a+c\,z^2}==\frac{d\,q+a\,e}{2\,a\,c}\,\frac{q+c\,z}{a+c\,z^2}+\frac{d\,q-a\,e}{2\,a\,c}\,\frac{q-c\,z}{a+c\,z^2}$

Note: Resulting integrands are of the form $\frac{d+e\,x^2}{a+c\,x^4}$ where $c\,d^2-a\,e^2==0$.

Rule 1.2.2.3.4.1.1.2.2.1: If $c\,d^2+a\,e^2\neq 0\,\land\,c\,d^2-a\,e^2\neq 0\,\land\,-a\,c\not>0$, let $q\to\sqrt{a\,c}$, then

$$\int \frac{d+e\,x^2}{a+c\,x^4}\,dx\,\longrightarrow\,\frac{d\,q+a\,e}{2\,a\,c}\int \frac{q+c\,x^2}{a+c\,x^4}\,dx+\frac{d\,q-a\,e}{2\,a\,c}\int \frac{q-c\,x^2}{a+c\,x^4}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)/(a_+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[a*c,2]},
  (d*q+a*e)/(2*a*c)*Int[(q+c*x^2)/(a+c*x^4),x] + (d*q-a*e)/(2*a*c)*Int[(q-c*x^2)/(a+c*x^4),x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && NegQ[-a*c]
```

2: $\int \frac{d+e x^2}{a+b x^2+c x^4} \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land b^2-4ac \not> 0$

Derivation: Algebraic expansion

Basis: If $q \to \sqrt{\frac{a}{c}}$ and $r \to \sqrt{2q-\frac{b}{c}}$, then $\frac{d+e z^2}{a+b z^2+c z^4} == \frac{d r-(d-e q) z}{2 c q r (q-r z+z^2)} + \frac{d r+(d-e q) z}{2 c q r (q+r z+z^2)}$

Note: If $(a \mid b \mid c) \in \mathbb{R} \land b^2-4ac<0$, then $\frac{a}{c}>0$ and $2\sqrt{\frac{a}{c}}-\frac{b}{c}>0$.

Rule 1.2.2.3.4.1.1.2.2.2: If $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land b^2-4ac \not> 0$, let $q \to \sqrt{\frac{a}{c}}$ and $r \to \sqrt{2q-\frac{b}{c}}$, then

$$\int \frac{d+e x^2}{a+b x^2+c x^4} \, dx \to \frac{1}{2 c q r} \int \frac{d r-(d-e q) x}{q-r x+x^2} \, dx + \frac{1}{2 c q r} \int \frac{d r+(d-e q) x}{q+r x+x^2} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{q=Rt[a/c,2]},
  With[{r=Rt[2*q-b/c,2]},
  1/(2*c*q*r)*Int[(d*r-(d-e*q)*x)/(q-r*x+x^2),x] + 1/(2*c*q*r)*Int[(d*r+(d-e*q)*x)/(q+r*x+x^2),x]]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NegQ[b^2-4*a*c]
```

2: $\int \frac{(d+e x^2)^q}{a+b x^2+c x^4} \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.2.2.3.4.1.2: If $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q \in \mathbb{Z}$, then

$$\int \frac{(d+e x^2)^q}{a+b x^2+c x^4}\,dx \to \int \text{ExpandIntegrand}\left[\frac{(d+e x^2)^q}{a+b x^2+c x^4},\,x\right]dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^q/(a+b*x^2+c*x^4),x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IntegerQ[q]
```

```
Int[(d_+e_.*x_^2)^q_/(a_+c_.*x_^4),x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^q/(a+c*x^4),x],x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && IntegerQ[q]
```

2. $\int \frac{(d+e x^2)^q}{a+b x^2+c x^4}\,dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q \notin \mathbb{Z}$

1: $\int \frac{(d+e x^2)^q}{a+b x^2+c x^4}\,dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q \notin \mathbb{Z} \land q<-1$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b z+c z^2} == \frac{e^2}{c d^2-b d e+a e^2} + \frac{(d+e z)\,(c d-b e-c e z)}{(c d^2-b d e+a e^2)\,(a+b z+c z^2)}$

Rule 1.2.2.3.4.2.1: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q \notin \mathbb{Z} \land q<-1$, then

$$\int \frac{(d+e x^2)^q}{a+b x^2+c x^4}\,dx \to \frac{e^2}{c d^2-b d e+a e^2}\int (d+e x^2)^q\,dx + \frac{1}{c d^2-b d e+a e^2}\int \frac{(d+e x^2)^{q+1}\,(c d-b e-c e x^2)}{a+b x^2+c x^4}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  e^2/(c*d^2-b*d*e+a*e^2)*Int[(d+e*x^2)^q,x] +
  1/(c*d^2-b*d*e+a*e^2)*Int[(d+e*x^2)^(q+1)*(c*d-b*e-c*e*x^2)/(a+b*x^2+c*x^4),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[q]] && LtQ[q,-1]
```

```
Int[(d_+e_.*x_^2)^q_/(a_+c_.*x_^4),x_Symbol] :=
  e^2/(c*d^2+a*e^2)*Int[(d+e*x^2)^q,x] +
  c/(c*d^2+a*e^2)*Int[(d+e*x^2)^(q+1)*(d-e*x^2)/(a+c*x^4),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[q]] && LtQ[q,-1]
```

2: $\int \frac{(d+e x^2)^q}{a+b x^2+c x^4} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q \notin \mathbb{Z} \land q \nless -1$

Derivation: Algebraic expansion

- Basis: If $r = \sqrt{b^2-4 a c}$, then $\frac{1}{a+b z+c z^2} = \frac{2 c}{r (b-r+2 c z)} - \frac{2 c}{r (b+r+2 c z)}$

Rule 1.2.2.3.4.2.2: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land q \notin \mathbb{Z} \land q \nless -1$, then

$$\int \frac{(d+e x^2)^q}{a+b x^2+c x^4} \, dx \to \frac{2 c}{r} \int \frac{(d+e x^2)^q}{b-r+2 c x^2} \, dx - \frac{2 c}{r} \int \frac{(d+e x^2)^q}{b+r+2 c x^2} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_/(a_+b_.*x_^2+c_.*x_^4),x_Symbol] :=
  With[{r=Rt[b^2-4*a*c,2]},
  2*c/r*Int[(d+e*x^2)^q/(b-r+2*c*x^2),x] - 2*c/r*Int[(d+e*x^2)^q/(b+r+2*c*x^2),x]] /;
FreeQ[{a,b,c,d,e,q},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && Not[IntegerQ[q]]
```

```
Int[(d_+e_.*x_^2)^q_/(a_+c_.*x_^4),x_Symbol] :=
  With[{r=Rt[-a*c,2]},
  -c/(2*r)*Int[(d+e*x^2)^q/(r-c*x^2),x] - c/(2*r)*Int[(d+e*x^2)^q/(r+c*x^2),x]] /;
FreeQ[{a,c,d,e,q},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[q]]
```

5. $\int (d+e x^2)\,(a+b x^2+c x^4)^p\,dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$

1: $\int (d+e x^2)\,(a+b x^2+c x^4)^p\,dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p>0$

Derivation: Trinomial recurrence 1b with $m=0$

Rule 1.2.2.3.5.1: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p>0$, then

$$\int (d+e x^2)\,(a+b x^2+c x^4)^p\,dx \longrightarrow$$

$$\frac{x\,\left(2 b e p+c d\,(4 p+3)+c e\,(4 p+1)\,x^2\right)\,(a+b x^2+c x^4)^p}{c\,(4 p+1)\,(4 p+3)}+$$

$$\frac{2 p}{c\,(4 p+1)\,(4 p+3)}\int \left(2 a c d\,(4 p+3)-a b e+\left(2 a c e\,(4 p+1)+b c d\,(4 p+3)-b^2 e\,(2 p+1)\right)\,x^2\right)\,(a+b x^2+c x^4)^{p-1}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  x*(2*b*e*p+c*d*(4*p+3)+c*e*(4*p+1)*x^2)*(a+b*x^2+c*x^4)^p/(c*(4*p+1)*(4*p+3)) +
  2*p/(c*(4*p+1)*(4*p+3))*Int[Simp[2*a*c*d*(4*p+3)-a*b*e+(2*a*c*e*(4*p+1)+b*c*d*(4*p+3)-b^2*e*(2*p+1))*x^2,x]*
    (a+b*x^2+c*x^4)^(p-1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && GtQ[p,0] && FractionQ[p] && IntegerQ[2*p]
```

```
Int[(d_+e_.*x_^2)*(a_+c_.*x_^4)^p_,x_Symbol] :=
  x*(d*(4*p+3)+e*(4*p+1)*x^2)*(a+c*x^4)^p/((4*p+1)*(4*p+3)) +
  2*p/((4*p+1)*(4*p+3))*Int[Simp[2*a*d*(4*p+3)+(2*a*e*(4*p+1))*x^2,x]*(a+c*x^4)^(p-1),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && GtQ[p,0] && FractionQ[p] && IntegerQ[2*p]
```

2: $\int \left(d+e x^2\right) \left(a+b x^2+c x^4\right)^p dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p<-1$

Derivation: Trinomial recurrence 2b with $m = 0$

Rule 1.2.2.3.5.2: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p<-1$, then

$$\int \left(d+e x^2\right) \left(a+b x^2+c x^4\right)^p dx \to$$
$$\frac{x \left(a b e-d \left(b^2-2 a c\right)-c \left(b d-2 a e\right) x^2\right) \left(a+b x^2+c x^4\right)^{p+1}}{2 a (p+1) \left(b^2-4 a c\right)}+$$
$$\frac{1}{2 a (p+1) \left(b^2-4 a c\right)} \int \left((2 p+3) d b^2-a b e-2 a c d (4 p+5)+(4 p+7) \left(d b-2 a e\right) c x^2\right) \left(a+b x^2+c x^4\right)^{p+1} dx$$

Program code:

```
Int[(d_+e_.*x_^2)*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  x*(a*b*e-d*(b^2-2*a*c)-c*(b*d-2*a*e)*x^2)*(a+b*x^2+c*x^4)^(p+1)/(2*a*(p+1)*(b^2-4*a*c)) +
  1/(2*a*(p+1)*(b^2-4*a*c))*Int[Simp[(2*p+3)*d*b^2-a*b*e-2*a*c*d*(4*p+5)+(4*p+7)*(d*b-2*a*e)*c*x^2,x]*
    (a+b*x^2+c*x^4)^(p+1),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && LtQ[p,-1] && IntegerQ[2*p]
```

```
Int[(d_+e_.*x_^2)*(a_+c_.*x_^4)^p_,x_Symbol] :=
  -x*(d+e*x^2)*(a+c*x^4)^(p+1)/(4*a*(p+1)) +
  1/(4*a*(p+1))*Int[Simp[d*(4*p+5)+e*(4*p+7)*x^2,x]*(a+c*x^4)^(p+1),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && LtQ[p,-1] && IntegerQ[2*p]
```

3. $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$

1. $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} dx$ when $b^2-4 a c>0$

1: $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c>0 \wedge c<0$

Derivation: Algebraic expansion

Basis: If $b^2-4 a c>0 \wedge c<0$, let $q \to \sqrt{b^2-4 a c}$, then
$\sqrt{a+b x^2+c x^4} == \frac{1}{2\sqrt{-c}} \sqrt{b+q+2 c x^2}\ \sqrt{-b+q-2 c x^2}$

- Rule 1.2.2.3.5.3.1.1: If $b^2-4 a c>0 \wedge c<0$, let $q \to \sqrt{b^2-4 a c}$, then

$$\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}}\,dx \longrightarrow 2\sqrt{-c} \int \frac{d+e x^2}{\sqrt{b+q+2 c x^2}\ \sqrt{-b+q-2 c x^2}}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*Sqrt[-c]*Int[(d+e*x^2)/(Sqrt[b+q+2*c*x^2]*Sqrt[-b+q-2*c*x^2]),x]] /;
FreeQ[{a,b,c,d,e},x] && GtQ[b^2-4*a*c,0] && LtQ[c,0]
```

```
Int[(d_+e_.*x_^2)/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  Sqrt[-c]*Int[(d+e*x^2)/(Sqrt[q+c*x^2]*Sqrt[q-c*x^2]),x]] /;
FreeQ[{a,c,d,e},x] && GtQ[a,0] && LtQ[c,0]
```

2. $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \wedge c \not< 0$

1. $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \wedge \frac{c}{a}>0 \wedge \frac{b}{a}<0$

1: $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \wedge \frac{c}{a}>0 \wedge \frac{b}{a}<0 \wedge e+d\sqrt{\frac{c}{a}}==0$

Reference: G&R 3.165.10

Rule 1.2.2.3.5.3.1.2.1.1: If $b^2-4 a c>0 \wedge \frac{c}{a}>0 \wedge \frac{b}{a}<0$, let $q \to \left(\frac{c}{a}\right)^{\frac{1}{4}}$, if $e+d q^2==0$, then

$$\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow -\frac{d x \sqrt{a+b x^2+c x^4}}{a \left(1+q^2 x^2\right)}+2 d \int \frac{\sqrt{a+b x^2+c x^4}}{a+2 a q^2 x^2+c x^4} \, dx$$

$$\longrightarrow -\frac{d x \sqrt{a+b x^2+c x^4}}{a \left(1+q^2 x^2\right)}+\frac{d \left(1+q^2 x^2\right) \sqrt{\frac{a+b x^2+c x^4}{a \left(1+q^2 x^2\right)^2}}}{q \sqrt{a+b x^2+c x^4}} \text{EllipticE}\left[2 \text{ArcTan}[q x], \frac{1}{2}-\frac{b q^2}{4 c}\right]$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[c/a,4]},
  -d*x*Sqrt[a+b*x^2+c*x^4]/(a*(1+q^2*x^2)) +
   d*(1+q^2*x^2)*Sqrt[(a+b*x^2+c*x^4)/(a*(1+q^2*x^2)^2)]/(q*Sqrt[a+b*x^2+c*x^4])*EllipticE[2*ArcTan[q*x],1/2-b*q^2/(4*c)] /;
 EqQ[e+d*q^2,0]] /;
FreeQ[{a,b,c,d,e},x] && GtQ[b^2-4*a*c,0] && GtQ[c/a,0] && LtQ[b/a,0]
```

2: $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land \frac{c}{a}>0 \land \frac{b}{a}<0 \land e+d \sqrt{\frac{c}{a}} \neq 0$

Derivation: Algebraic expansion

- Rule 1.2.2.3.5.3.1.2.1.2: If $b^2-4 a c>0 \land \frac{c}{a}>0 \land \frac{b}{a}<0$, let $q \to \sqrt{\frac{c}{a}}$, if $e+d q \neq 0$, then

$$\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx \to \frac{e+d q}{q} \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx-\frac{e}{q} \int \frac{1-q x^2}{\sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[c/a,2]},
  (e+d*q)/q*Int[1/Sqrt[a+b*x^2+c*x^4],x] - e/q*Int[(1-q*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
 NeQ[e+d*q,0]] /;
FreeQ[{a,b,c,d,e},x] && GtQ[b^2-4*a*c,0] && GtQ[c/a,0] && LtQ[b/a,0]
```

2. $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land a<0 \land c>0$

1: $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land a<0 \land c>0 \land 2 c d-e \left(b-\sqrt{b^2-4 a c}\right)==0$

Reference: G&R 3.153.2+

- Rule 1.2.2.3.5.3.1.2.2.1: If $b^2-4 a c>0 \land a<0 \land c>0$, let $q \to \sqrt{b^2-4 a c}$, if $2 c d-e (b-q)==0$, then

$$\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx \to \frac{e x \left(b+q+2 c x^2\right)}{2 c \sqrt{a+b x^2+c x^4}}-\frac{e q}{2 c} \int \frac{2 a+(b-q) x^2}{\left(a+b x^2+c x^4\right)^{3/2}} \, dx$$

$$\to \frac{e\,x\,\left(b+q+2\,c\,x^2\right)}{2\,c\,\sqrt{a+b\,x^2+c\,x^4}} - \frac{e\,q\,\sqrt{\frac{2\,a+(b-q)\,x^2}{2\,a+(b+q)\,x^2}}\,\sqrt{\frac{2\,a+(b+q)\,x^2}{q}}}{2\,c\,\sqrt{a+b\,x^2+c\,x^4}\,\sqrt{\frac{a}{2\,a+(b+q)\,x^2}}}\,\text{EllipticE}\left[\text{ArcSin}\left[\frac{x}{\sqrt{\frac{2\,a+(b+q)\,x^2}{2\,q}}}\right],\frac{b+q}{2\,q}\right]$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  e*x*(b+q+2*c*x^2)/(2*c*Sqrt[a+b*x^2+c*x^4]) -
  e*q*Sqrt[(2*a+(b-q)*x^2)/(2*a+(b+q)*x^2)]*Sqrt[(2*a+(b+q)*x^2)/q]/(2*c*Sqrt[a+b*x^2+c*x^4]*Sqrt[a/(2*a+(b+q)*x^2)])*
     EllipticE[ArcSin[x/Sqrt[(2*a+(b+q)*x^2)/(2*q)]],(b+q)/(2*q)] /;
 EqQ[2*c*d-e*(b-q),0]] /;
FreeQ[{a,b,c,d,e},x] && GtQ[b^2-4*a*c,0] && LtQ[a,0] && GtQ[c,0]
```

```
Int[(d_+e_.*x_^2)/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  e*x*(q+c*x^2)/(c*Sqrt[a+c*x^4]) -
  Sqrt[2]*e*q*Sqrt[-a+q*x^2]*Sqrt[(a+q*x^2)/q]/(Sqrt[-a]*c*Sqrt[a+c*x^4])*
     EllipticE[ArcSin[x/Sqrt[(a+q*x^2)/(2*q)]],1/2] /;
 EqQ[c*d+e*q,0] && IntegerQ[q]] /;
FreeQ[{a,c,d,e},x] && LtQ[a,0] && GtQ[c,0]
```

```
Int[(d_+e_.*x_^2)/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  e*x*(q+c*x^2)/(c*Sqrt[a+c*x^4]) -
  Sqrt[2]*e*q*Sqrt[(a-q*x^2)/(a+q*x^2)]*Sqrt[(a+q*x^2)/q]/(c*Sqrt[a+c*x^4]*Sqrt[a/(a+q*x^2)])*
     EllipticE[ArcSin[x/Sqrt[(a+q*x^2)/(2*q)]],1/2] /;
 EqQ[c*d+e*q,0]] /;
FreeQ[{a,c,d,e},x] && LtQ[a,0] && GtQ[c,0]
```

2: $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land a<0 \land c>0 \land 2 c d-e \left(b-\sqrt{b^2-4 a c}\right) \neq 0$

Derivation: Algebraic expansion

- Rule 1.2.2.3.5.3.1.2.2.2: If $b^2-4 a c>0 \land a<0 \land c>0$, let $q \to \sqrt{b^2-4 a c}$, if $2 c d-e (b-q) \neq 0$, then

$$\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow \frac{2 c d-e (b-q)}{2 c} \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx+\frac{e}{2 c} \int \frac{b-q+2 c x^2}{\sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (2*c*d-e*(b-q))/(2*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + e/(2*c)*Int[(b-q+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
 NeQ[2*c*d-e*(b-q),0]] /;
FreeQ[{a,b,c,d,e},x] && GtQ[b^2-4*a*c,0] && LtQ[a,0] && GtQ[c,0]
```

```
Int[(d_+e_.*x_^2)/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  (c*d+e*q)/c*Int[1/Sqrt[a+c*x^4],x] - e/c*Int[(q-c*x^2)/Sqrt[a+c*x^4],x] /;
 NeQ[c*d+e*q,0]] /;
FreeQ[{a,c,d,e},x] && LtQ[a,0] && GtQ[c,0]
```

3: $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land \frac{b\pm\sqrt{b^2-4 a c}}{a}>0$

Derivation: Algebraic expansion

- Rule 1.2.2.3.5.3.1.2.3: If $b^2-4 a c>0$, let $q \to \sqrt{b^2-4 a c}$, if $\frac{b\pm q}{a}>0$, then

$$\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow d \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx + e \int \frac{x^2}{\sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  d*Int[1/Sqrt[a+b*x^2+c*x^4],x] + e*Int[x^2/Sqrt[a+b*x^2+c*x^4],x] /;
 PosQ[(b+q)/a] || PosQ[(b-q)/a]] /;
FreeQ[{a,b,c,d,e},x] && GtQ[b^2-4*a*c,0]
```

```
Int[(d_+e_.*x_^2)/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  d*Int[1/Sqrt[a+c*x^4],x] + e*Int[x^2/Sqrt[a+c*x^4],x] /;
FreeQ[{a,c,d,e},x] && GtQ[-a*c,0]
```

4. $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land \frac{b\pm\sqrt{b^2-4 a c}}{a} \ngtr 0$

1. $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land \frac{b+\sqrt{b^2-4 a c}}{a} \ngtr 0$

1: $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \land \frac{b+\sqrt{b^2-4 a c}}{a} \ngtr 0 \land 2 c d-e (b+q) == 0$

Reference: G&R 3.153.5+

Rule 1.2.2.3.5.3.1.2.4.1.1: If $b^2-4 a c>0$, let $q \to \sqrt{b^2-4 a c}$, if $\frac{b+q}{a} \ngtr 0 \land 2 c d-e (b+q) == 0$ then

$$\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx \to -\frac{a e \sqrt{-\frac{b+q}{2 a}} \sqrt{1+\frac{(b+q) x^2}{2 a}} \sqrt{1+\frac{(b-q) x^2}{2 a}}}{c \sqrt{a+b x^2+c x^4}} \text{EllipticE}\left[\text{ArcSin}\left[\sqrt{-\frac{b+q}{2 a}} x\right], \frac{b-q}{b+q}\right]$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  -a*e*Rt[-(b+q)/(2*a),2]*Sqrt[1+(b+q)*x^2/(2*a)]*Sqrt[1+(b-q)*x^2/(2*a)]/(c*Sqrt[a+b*x^2+c*x^4])*
    EllipticE[ArcSin[Rt[-(b+q)/(2*a),2]*x],(b-q)/(b+q)] /;
 NegQ[(b+q)/a] && EqQ[2*c*d-e*(b+q),0] && Not[SimplerSqrtQ[-(b-q)/(2*a),-(b+q)/(2*a)]]] /;
FreeQ[{a,b,c,d,e},x] && GtQ[b^2-4*a*c,0]
```

2: $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c>0 \wedge \frac{b+\sqrt{b^2-4 a c}}{a} \not> 0 \wedge 2 c d-e \left(b+q\right) \neq 0$

Derivation: Algebraic expansion

- Rule 1.2.2.3.5.3.1.2.4.1.2: If $b^2-4 a c>0$, let $q \to \sqrt{b^2-4 a c}$, if $\frac{b+q}{a} \not> 0 \wedge 2 c d-e \left(b+q\right) \neq 0$ then

$$\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow \frac{2 c d-e \left(b+q\right)}{2 c} \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx+\frac{e}{2 c} \int \frac{b+q+2 c x^2}{\sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (2*c*d-e*(b+q))/(2*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + e/(2*c)*Int[(b+q+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
 NegQ[(b+q)/a] && NeQ[2*c*d-e*(b+q),0] && Not[SimplerSqrtQ[-(b-q)/(2*a),-(b+q)/(2*a)]]] /;
FreeQ[{a,b,c,d,e},x] && GtQ[b^2-4*a*c,0]
```

2. $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac>0 \land \frac{b-\sqrt{b^2-4ac}}{a} \ngtr 0$

1: $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac>0 \land \frac{b-\sqrt{b^2-4ac}}{a} \ngtr 0 \land 2cd-e\,(b-q)==0$

Reference: G&R 3.153.5-

- Rule 1.2.2.3.5.3.1.2.4.2.1: If $b^2-4ac>0$, let $q \to \sqrt{b^2-4ac}$, if $\frac{b-q}{a} \ngtr 0 \land 2cd-e\,(b-q)==0$ then

$$\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}}\,dx \to -\frac{a e \sqrt{-\frac{b-q}{2a}}\,\sqrt{1+\frac{(b-q)\,x^2}{2a}}\,\sqrt{1+\frac{(b+q)\,x^2}{2a}}}{c\,\sqrt{a+b x^2+c x^4}}\,\text{EllipticE}\left[\text{ArcSin}\left[\sqrt{-\frac{b-q}{2a}}\,x\right],\frac{b+q}{b-q}\right]$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  -a*e*Rt[-(b-q)/(2*a),2]*Sqrt[1+(b-q)*x^2/(2*a)]*Sqrt[1+(b+q)*x^2/(2*a)]/(c*Sqrt[a+b*x^2+c*x^4])*
    EllipticE[ArcSin[Rt[-(b-q)/(2*a),2]*x],(b+q)/(b-q)] /;
 NegQ[(b-q)/a] && EqQ[2*c*d-e*(b-q),0]] /;
FreeQ[{a,b,c,d,e},x] && GtQ[b^2-4*a*c,0]
```

2: $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac>0 \land \frac{b-\sqrt{b^2-4ac}}{a} \ngtr 0 \land 2cd-e\,(b-q)\neq 0$

Derivation: Algebraic expansion

- Rule 1.2.2.3.5.3.1.2.4.2.2: If $b^2-4ac>0$, let $q \to \sqrt{b^2-4ac}$, if $\frac{b-q}{a} \ngtr 0 \land 2cd-e\,(b-q)\neq 0$ then

$$\int \frac{d + e\,x^2}{\sqrt{a + b\,x^2 + c\,x^4}}\,dx \longrightarrow \frac{2\,c\,d - e\,(b - q)}{2\,c}\int \frac{1}{\sqrt{a + b\,x^2 + c\,x^4}}\,dx + \frac{e}{2\,c}\int \frac{b - q + 2\,c\,x^2}{\sqrt{a + b\,x^2 + c\,x^4}}\,dx$$

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  (2*c*d-e*(b-q))/(2*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + e/(2*c)*Int[(b-q+2*c*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
 NegQ[(b-q)/a] && NeQ[2*c*d-e*(b-q),0]] /;
FreeQ[{a,b,c,d,e},x] && GtQ[b^2-4*a*c,0]
```

2. $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}}\, dx$ when $b^2-4 a c \ngtr 0$

1. $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}}\, dx$ when $b^2-4 a c \neq 0 \wedge \frac{c}{a}>0$

1: $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}}\, dx$ when $b^2-4 a c \neq 0 \wedge \frac{c}{a}>0 \wedge e+d\sqrt{\frac{c}{a}} == 0$

Reference: G&R 3.165.10

Rule 1.2.2.3.5.3.2.1.1: If $b^2-4 a c \neq 0 \wedge \frac{c}{a}>0$, let $q=\left(\frac{c}{a}\right)^{\frac{1}{4}}$, if $e+d q^2 == 0$, then

$$\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}}\, dx \longrightarrow -\frac{d x \sqrt{a+b x^2+c x^4}}{a\left(1+q^2 x^2\right)}+2 d \int \frac{\sqrt{a+b x^2+c x^4}}{a+2 a q^2 x^2+c x^4}\, dx$$

$$\longrightarrow -\frac{d x \sqrt{a+b x^2+c x^4}}{a\left(1+q^2 x^2\right)}+\frac{d\left(1+q^2 x^2\right)\sqrt{\frac{a+b x^2+c x^4}{a\left(1+q^2 x^2\right)^2}}}{q \sqrt{a+b x^2+c x^4}}\, \text{EllipticE}\left[2 \text{ArcTan}[q x], \frac{1}{2}-\frac{b q^2}{4 c}\right]$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[c/a,4]},
  -d*x*Sqrt[a+b*x^2+c*x^4]/(a*(1+q^2*x^2)) +
   d*(1+q^2*x^2)*Sqrt[(a+b*x^2+c*x^4)/(a*(1+q^2*x^2)^2)]/(q*Sqrt[a+b*x^2+c*x^4])*EllipticE[2*ArcTan[q*x],1/2-b*q^2/(4*c)] /;
 EqQ[e+d*q^2,0]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && PosQ[c/a]
```

```
Int[(d_+e_.*x_^2)/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[c/a,4]},
  -d*x*Sqrt[a+c*x^4]/(a*(1+q^2*x^2)) +
   d*(1+q^2*x^2)*Sqrt[(a+c*x^4)/(a*(1+q^2*x^2)^2)]/(q*Sqrt[a+c*x^4])*EllipticE[2*ArcTan[q*x],1/2] /;
 EqQ[e+d*q^2,0]] /;
FreeQ[{a,c,d,e},x] && PosQ[c/a]
```

2: $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \land \frac{c}{a}>0 \land e+d \sqrt{\frac{c}{a}} \neq 0$

Derivation: Algebraic expansion

Rule 1.2.2.3.5.3.2.1.2: If $b^2-4 a c \neq 0 \land \frac{c}{a}>0$, let $q \to \sqrt{\frac{c}{a}}$, if $e+d q \neq 0$, then

$$\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx \to \frac{e+d q}{q} \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx-\frac{e}{q} \int \frac{1-q x^2}{\sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[c/a,2]},
  (e+d*q)/q*Int[1/Sqrt[a+b*x^2+c*x^4],x] - e/q*Int[(1-q*x^2)/Sqrt[a+b*x^2+c*x^4],x] /;
 NeQ[e+d*q,0]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && PosQ[c/a]
```

```
Int[(d_+e_.*x_^2)/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[c/a,2]},
  (e+d*q)/q*Int[1/Sqrt[a+c*x^4],x] - e/q*Int[(1-q*x^2)/Sqrt[a+c*x^4],x] /;
 NeQ[e+d*q,0]] /;
FreeQ[{a,c,d,e},x] && PosQ[c/a]
```

2. $\int \frac{d+e x^2}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \land \frac{c}{a} \not> 0$

1. $\int \frac{d+e x^2}{\sqrt{a+c x^4}} \, dx$ when $\frac{c}{a} \not> 0$

1. $\int \frac{d+e x^2}{\sqrt{a+c x^4}} \, dx$ when $\frac{c}{a} \not> 0 \land c d^2+a e^2 == 0$

1: $\int \frac{d+e x^2}{\sqrt{a+c x^4}} \, dx$ when $\frac{c}{a} \not> 0 \land c d^2 + a e^2 == 0 \land a > 0$

Derivation: Algebraic expansion

■ Basis: If $c d^2 + a e^2 == 0 \land a > 0$, then $\frac{d+e x^2}{\sqrt{a+c x^4}} == \frac{d \sqrt{1+\frac{e x^2}{d}}}{\sqrt{a} \sqrt{1-\frac{e x^2}{d}}}$

Rule 1.2.2.3.5.3.2.2.1.1.1: If $\frac{c}{a} \not> 0 \land c d^2 + a e^2 == 0 \land a > 0$, then

$$\int \frac{d+e x^2}{\sqrt{a+c x^4}} \, dx \longrightarrow \frac{d}{\sqrt{a}} \int \frac{\sqrt{1+\frac{e x^2}{d}}}{\sqrt{1-\frac{e x^2}{d}}} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  d/Sqrt[a]*Int[Sqrt[1+e*x^2/d]/Sqrt[1-e*x^2/d],x] /;
FreeQ[{a,c,d,e},x] && NegQ[c/a] && EqQ[c*d^2+a*e^2,0] && GtQ[a,0]
```

2: $\int \frac{d+e x^2}{\sqrt{a+c x^4}} \, dx$ when $\frac{c}{a} \not> 0 \land c d^2 + a e^2 == 0 \land a \not> 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{1+\frac{c x^4}{a}}}{\sqrt{a+c x^4}} == 0$

Rule 1.2.2.3.5.3.2.2.1.1.2: If $\frac{c}{a} \not> 0 \land c d^2 + a e^2 == 0 \land a \not> 0$, then

$$\int \frac{d + e\,x^2}{\sqrt{a + c\,x^4}}\,dx \longrightarrow \frac{\sqrt{1 + \frac{c\,x^4}{a}}}{\sqrt{a + c\,x^4}} \int \frac{d + e\,x^2}{\sqrt{1 + \frac{c\,x^4}{a}}}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  Sqrt[1+c*x^4/a]/Sqrt[a+c*x^4]*Int[(d+e*x^2)/Sqrt[1+c*x^4/a],x] /;
FreeQ[{a,c,d,e},x] && NegQ[c/a] && EqQ[c*d^2+a*e^2,0] && Not[GtQ[a,0]]
```

2: $\int \frac{d + e\,x^2}{\sqrt{a + c\,x^4}}\,dx$ when $\frac{c}{a} \ngtr 0 \land c\,d^2 + a\,e^2 \neq 0$

Derivation: Algebraic expansion

Basis: $d + e\,x^2 == \frac{d\,q - e}{q} + \frac{e\,(1 + q\,x^2)}{q}$

Rule 1.2.2.3.5.3.2.2.1.2: If $\frac{c}{a} \ngtr 0 \land c\,d^2 + a\,e^2 \neq 0$, let $q \to \sqrt{-\frac{c}{a}}$, then

$$\int \frac{d + e\,x^2}{\sqrt{a + c\,x^4}}\,dx \longrightarrow \frac{d\,q - e}{q} \int \frac{1}{\sqrt{a + c\,x^4}}\,dx + \frac{e}{q} \int \frac{1 + q\,x^2}{\sqrt{a + c\,x^4}}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[-c/a,2]},
  (d*q-e)/q*Int[1/Sqrt[a+c*x^4],x] + e/q*Int[(1+q*x^2)/Sqrt[a+c*x^4],x]] /;
FreeQ[{a,c,d,e},x] && NegQ[c/a] && NeQ[c*d^2+a*e^2,0]
```

2: $\int \frac{d+e\,x^2}{\sqrt{a+b\,x^2+c\,x^4}}\,dx$ when $b^2-4\,a\,c\neq 0\,\wedge\,\frac{c}{a}\not>0$

Derivation: Piecewise constant extraction

Basis: If $q\to\sqrt{b^2-4\,a\,c}$, then $\partial_x\frac{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}{\sqrt{a+b\,x^2+c\,x^4}}=0$

- Rule 1.2.2.3.5.3.2.2.2: If $b^2-4\,a\,c\neq 0\,\wedge\,\frac{c}{a}\not>0$, let $q\to\sqrt{b^2-4\,a\,c}$, then

$$\int\frac{d+e\,x^2}{\sqrt{a+b\,x^2+c\,x^4}}\,dx\,\to\,\frac{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}{\sqrt{a+b\,x^2+c\,x^4}}\int\frac{d+e\,x^2}{\sqrt{1+\frac{2\,c\,x^2}{b-q}}\,\sqrt{1+\frac{2\,c\,x^2}{b+q}}}\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]/Sqrt[a+b*x^2+c*x^4]*
    Int[(d+e*x^2)/(Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NegQ[c/a]
```

4: $\int (d + e x^2) (a + b x^2 + c x^4)^p \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0$

Derivation: Algebraic expansion

Rule 1.2.2.3.5.4: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0$, then

$$\int (d + e x^2) (a + b x^2 + c x^4)^p \, dx \to \int \text{ExpandIntegrand}\left[(d + e x^2) (a + b x^2 + c x^4)^p, x\right] dx$$

Program code:

```
Int[(d_+e_.*x_^2)*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)*(a+b*x^2+c*x^4)^p,x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[(d_+e_.*x_^2)*(a_+c_.*x_^4)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)*(a+c*x^4)^p,x],x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0]
```

6. $\int (d + e x^2)^q (a + b x^2 + c x^4)^p \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0 \land q - 1 \in \mathbb{Z}^+$

x: $\int \frac{(d + e x^2)^2}{\sqrt{a + b x^2 + c x^4}} \, dx$ when $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0$

Rule 1.2.2.3.6.x: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0$, then

$$\int \frac{(d + e x^2)^2}{\sqrt{a + b x^2 + c x^4}} \, dx \longrightarrow$$

$$\frac{e^2 x \sqrt{a + b x^2 + c x^4}}{3 c} + \frac{2 (3 c d - b e)}{3 c} \int \frac{d + e x^2}{\sqrt{a + b x^2 + c x^4}} \, dx - \frac{3 c d^2 - 2 b d e + a e^2}{3 c} \int \frac{1}{\sqrt{a + b x^2 + c x^4}} \, dx$$

Program code:

```
(* Int[(d_+e_.*x_^2)^2/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  e^2*x*Sqrt[a+b*x^2+c*x^4]/(3*c) +
  2*(3*c*d-b*e)/(3*c)*Int[(d+e*x^2)/Sqrt[a+b*x^2+c*x^4],x] -
  (3*c*d^2-2*b*d*e+a*e^2)/(3*c)*Int[1/Sqrt[a+b*x^2+c*x^4],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] *)
```

```
(* Int[(d_+e_.*x_^2)^2/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  e^2*x*Sqrt[a+c*x^4]/(3*c) +
  2*d*Int[(d+e*x^2)/Sqrt[a+c*x^4],x] -
  (3*c*d^2+a*e^2)/(3*c)*Int[1/Sqrt[a+c*x^4],x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] *)
```

x: $\int \frac{(d+e x^2)^q}{\sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \land q-2 \in \mathbb{Z}^+$

Rule 1.2.2.3.6.x: If $b^2-4 a c \neq 0 \land q-2 \in \mathbb{Z}^+$, then

$$\int \frac{(d+e x^2)^q}{\sqrt{a+b x^2+c x^4}} \, dx \longrightarrow$$

$$\frac{e^2 x (d+e x^2)^{q-2} \sqrt{a+b x^2+c x^4}}{c (2 q-1)}+\frac{2 (q-1) (3 c d-b e)}{c (2 q-1)} \int \frac{(d+e x^2)^{q-1}}{\sqrt{a+b x^2+c x^4}} \, dx -$$
$$\frac{(2 q-3) (3 c d^2-2 b d e+a e^2)}{c (2 q-1)} \int \frac{(d+e x^2)^{q-2}}{\sqrt{a+b x^2+c x^4}} \, dx+\frac{2 d (q-2) (c d^2-b d e+a e^2)}{c (2 q-1)} \int \frac{(d+e x^2)^{q-3}}{\sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
(* Int[(d_+e_.*x_^2)^q_/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  e^2*x*(d+e*x^2)^(q-2)*Sqrt[a+b*x^2+c*x^4]/(c*(2*q-1)) +
  2*(q-1)*(3*c*d-b*e)/(c*(2*q-1))*Int[(d+e*x^2)^(q-1)/Sqrt[a+b*x^2+c*x^4],x] -
  (2*q-3)*(3*c*d^2-2*b*d*e+a*e^2)/(c*(2*q-1))*Int[(d+e*x^2)^(q-2)/Sqrt[a+b*x^2+c*x^4],x] +
  2*d*(q-2)*(c*d^2-b*d*e+a*e^2)/(c*(2*q-1))*Int[(d+e*x^2)^(q-3)/Sqrt[a+b*x^2+c*x^4],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && IGtQ[q,2] *)
```

```
(* Int[(d_+e_.*x_^2)^q_/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  e^2*x*(d+e*x^2)^(q-2)*Sqrt[a+c*x^4]/(c*(2*q-1)) +
  6*d*(q-1)/(2*q-1)*Int[(d+e*x^2)^(q-1)/Sqrt[a+c*x^4],x] -
  (2*q-3)*(3*c*d^2+a*e^2)/(c*(2*q-1))*Int[(d+e*x^2)^(q-2)/Sqrt[a+c*x^4],x] +
  2*d*(q-2)*(c*d^2+a*e^2)/(c*(2*q-1))*Int[(d+e*x^2)^(q-3)/Sqrt[a+c*x^4],x] /;
FreeQ[{a,c,d,e},x] && IGtQ[q,2] *)
```

1: $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge q-1 \in \mathbb{Z}^+ \wedge p<-1$

Derivation: Algebraic expansion and trinomial recurrence 2b

Rule 1.2.2.3.6.1: If $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge q-1 \in \mathbb{Z}^+ \wedge p<-1$, let $Q_{q-2}[x^2] \to \text{PolynomialQuotient}[(d+e x^2)^q, a+b x^2+c x^4, x]$ and $f+g x^2 \to \text{PolynomialRemainder}[(d+e x^2)^q, a+b x^2+c x^4, x]$, then

$$\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \longrightarrow$$

$$\int (f+g x^2)(a+b x^2+c x^4)^p \, dx+\int Q_{q-2}[x^2](a+b x^2+c x^4)^{p+1} \, dx \longrightarrow$$

$$\frac{x (a+b x^2+c x^4)^{p+1} (a b g-f (b^2-2 a c)-c (b f-2 a g) x^2)}{2 a (p+1)(b^2-4 a c)}+\frac{1}{2 a (p+1)(b^2-4 a c)} \int (a+b x^2+c x^4)^{p+1} \cdot$$
$$(2 a (p+1)(b^2-4 a c) Q_{q-2}[x^2]+b^2 f (2 p+3)-2 a c f (4 p+5)-a b g+c (4 p+7)(b f-2 a g) x^2 \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  With[{f=Coeff[PolynomialRemainder[(d+e*x^2)^q,a+b*x^2+c*x^4,x],x,0],
        g=Coeff[PolynomialRemainder[(d+e*x^2)^q,a+b*x^2+c*x^4,x],x,2]},
  x*(a+b*x^2+c*x^4)^(p+1)*(a*b*g-f*(b^2-2*a*c)-c*(b*f-2*a*g)*x^2)/(2*a*(p+1)*(b^2-4*a*c)) +
  1/(2*a*(p+1)*(b^2-4*a*c))*Int[(a+b*x^2+c*x^4)^(p+1)*
    ExpandToSum[2*a*(p+1)*(b^2-4*a*c)*PolynomialQuotient[(d+e*x^2)^q,a+b*x^2+c*x^4,x]+
      b^2*f*(2*p+3)-2*a*c*f*(4*p+5)-a*b*g+c*(4*p+7)*(b*f-2*a*g)*x^2,x],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[q,1] && LtQ[p,-1]
```

2: $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4ac \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge q-1 \in \mathbb{Z}^+ \wedge p \nless -1$

Derivation: Algebraic expansion and

Note: This rule reduces the degree of the polynomial factor $(d+e x^2)^q$ in the resulting integrand.

Rule: 1.2.2.3.6.2: If $b^2-4ac \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge q-1 \in \mathbb{Z}^+ \wedge p \nless -1$, then

$$\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \longrightarrow$$

$$\int \left((d+e x^2)^q - e^q x^{2q}\right) (a+b x^2+c x^4)^p \, dx + e^q \int x^{2q} (a+b x^2+c x^4)^p \, dx \longrightarrow$$

$$\frac{e^q x^{2q-3} (a+b x^2+c x^4)^{p+1}}{c (4p+2q+1)} + \frac{1}{c (4p+2q+1)} \int (a+b x^2+c x^4)^p \cdot$$
$$\left(c (4p+2q+1) (d+e x^2)^q - a (2q-3) e^q x^{2q-4} - b (2p+2q-1) e^q x^{2q-2} - c (4p+2q+1) e^q x^{2q}\right) dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  e^q*x^(2*q-3)*(a+b*x^2+c*x^4)^(p+1)/(c*(4*p+2*q+1)) +
  1/(c*(4*p+2*q+1))*Int[(a+b*x^2+c*x^4)^p*
    ExpandToSum[c*(4*p+2*q+1)*(d+e*x^2)^q-a*(2*q-3)*e^q*x^(2*q-4)-b*(2*p+2*q-1)*e^q*x^(2*q-2)-c*(4*p+2*q+1)*e^q*x^(2*q),x],x] /;
FreeQ[{a,b,c,d,e,p},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[q,1]
```

```
Int[(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_,x_Symbol] :=
  e^q*x^(2*q-3)*(a+c*x^4)^(p+1)/(c*(4*p+2*q+1)) +
  1/(c*(4*p+2*q+1))*Int[(a+c*x^4)^p*
    ExpandToSum[c*(4*p+2*q+1)*(d+e*x^2)^q-a*(2*q-3)*e^q*x^(2*q-4)-c*(4*p+2*q+1)*e^q*x^(2*q),x],x] /;
FreeQ[{a,c,d,e,p},x] && NeQ[c*d^2+a*e^2,0] && IGtQ[q,1]
```

7. $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p+\frac{1}{2} \in \mathbb{Z} \land q \in \mathbb{Z}^-$

1. $\int \frac{(a+b x^2+c x^4)^p}{d+e x^2} \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p+\frac{1}{2} \in \mathbb{Z}$

1: $\int \frac{(a+b x^2+c x^4)^p}{d+e x^2} \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p+\frac{1}{2} \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{a+b x^2+c x^4}{d+e x^2} = -\frac{c d-b e-c e x^2}{e^2}+\frac{c d^2-b d e+a e^2}{e^2 (d+e x^2)}$

Rule 1.2.2.3.7.1: If $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p+\frac{1}{2} \in \mathbb{Z}^+$, then

$$\int \frac{(a+b x^2+c x^4)^p}{d+e x^2} \, dx \to -\frac{1}{e^2} \int (c d-b e-c e x^2) (a+b x^2+c x^4)^{p-1} \, dx+\frac{c d^2-b d e+a e^2}{e^2} \int \frac{(a+b x^2+c x^4)^{p-1}}{d+e x^2} \, dx$$

Program code:

```
Int[(a_+b_.*x_^2+c_.*x_^4)^p_/(d_+e_.*x_^2),x_Symbol] :=
  -1/e^2*Int[(c*d-b*e-c*e*x^2)*(a+b*x^2+c*x^4)^(p-1),x] +
  (c*d^2-b*d*e+a*e^2)/e^2*Int[(a+b*x^2+c*x^4)^(p-1)/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && IGtQ[p+1/2,0]
```

```
Int[(a_+c_.*x_^4)^p_/(d_+e_.*x_^2),x_Symbol] :=
  -1/e^2*Int[(c*d-c*e*x^2)*(a+c*x^4)^(p-1),x] +
  (c*d^2+a*e^2)/e^2*Int[(a+c*x^4)^(p-1)/(d+e*x^2),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && IGtQ[p+1/2,0]
```

2. $\int \frac{(a+b x^2+c x^4)^p}{d+e x^2} \, dx$ when $b^2-4ac \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land p-\frac{1}{2} \in \mathbb{Z}^-$

1. $\int \frac{1}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0$

1: $\int \frac{1}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge c d^2-a e^2 == 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{d+e x^2} == \frac{1}{2 d}+\frac{d-e x^2}{2 d (d+e x^2)}$

Rule 1.2.2.3.7.2.1.1: If $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0 \wedge c d^2-a e^2 == 0$, then

$$\int \frac{1}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx \to \frac{1}{2 d}\int \frac{1}{\sqrt{a+b x^2+c x^4}}\,dx+\frac{1}{2 d}\int \frac{d-e x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$$

Program code:

```
Int[1/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  1/(2*d)*Int[1/Sqrt[a+b*x^2+c*x^4],x] + 1/(2*d)*Int[(d-e*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && EqQ[c*d^2-a*e^2,0]
```

```
Int[1/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  1/(2*d)*Int[1/Sqrt[a+c*x^4],x] + 1/(2*d)*Int[(d-e*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && EqQ[c*d^2-a*e^2,0]
```

2. $\int \frac{1}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land c d^2-a e^2 \neq 0$

1. $\int \frac{1}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c > 0 \land c d^2-b d e+a e^2 \neq 0 \land c d^2-a e^2 \neq 0$

1: $\int \frac{1}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c > 0 \land c < 0$

Derivation: Algebraic expansion

Basis: If $b^2-4 a c > 0 \land c < 0$, let $q \to \sqrt{b^2-4 a c}$, then

$$\sqrt{a+b x^2+c x^4} == \frac{1}{2 \sqrt{-c}} \sqrt{b+q+2 c x^2} \sqrt{-b+q-2 c x^2}$$

Rule 1.2.2.3.7.2.1.2.1.1: If $b^2-4 a c > 0 \land c < 0$, let $q \to \sqrt{b^2-4 a c}$, then

$$\int \frac{1}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx \longrightarrow 2 \sqrt{-c} \int \frac{1}{\left(d+e x^2\right) \sqrt{b+q+2 c x^2} \sqrt{-b+q-2 c x^2}} \, dx$$

Program code:

```
Int[1/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*Sqrt[-c]*Int[1/((d+e*x^2)*Sqrt[b+q+2*c*x^2]*Sqrt[-b+q-2*c*x^2]),x]] /;
FreeQ[{a,b,c,d,e},x] && GtQ[b^2-4*a*c,0] && LtQ[c,0]
```

```
Int[1/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  Sqrt[-c]*Int[1/((d+e*x^2)*Sqrt[q+c*x^2]*Sqrt[q-c*x^2]),x]] /;
FreeQ[{a,c,d,e},x] && GtQ[a,0] && LtQ[c,0]
```

2: $\int \frac{1}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c>0 \land c \not< 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{d+e x^2} == \frac{2 c}{2 c d-e (b-q)} - \frac{e (b-q+2 c x^2)}{(2 c d-e (b-q)) (d+e x^2)}$

- Rule 1.2.2.3.7.2.1.2.1.2: If $b^2-4 a c>0 \land c \not< 0$, let $q \to \sqrt{b^2-4 a c}$, then

$$\int \frac{1}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx \to \frac{2 c}{2 c d-e (b-q)} \int \frac{1}{\sqrt{a+b x^2+c x^4}}\,dx - \frac{e}{2 c d-e (b-q)} \int \frac{b-q+2 c x^2}{(d+e x^2)\sqrt{a+b x^2+c x^4}}\,dx$$

Program code:

```
Int[1/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  2*c/(2*c*d-e*(b-q))*Int[1/Sqrt[a+b*x^2+c*x^4],x] - e/(2*c*d-e*(b-q))*Int[(b-q+2*c*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x]] /;
FreeQ[{a,b,c,d,e},x] && GtQ[b^2-4*a*c,0] && Not[LtQ[c,0]]
```

```
Int[1/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[-a*c,2]},
  c/(c*d+e*q)*Int[1/Sqrt[a+c*x^4],x] + e/(c*d+e*q)*Int[(q-c*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x]] /;
FreeQ[{a,c,d,e},x] && GtQ[-a*c,0] && Not[LtQ[c,0]]
```

2. $\int \frac{1}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land c d^2-a e^2 \neq 0$

1: $\int \frac{1}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land c d^2-a e^2 \neq 0 \land \frac{c}{a} > 0$

Derivation: Algebraic expansion

Rule 1.2.2.3.7.2.1.2.2.1: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land c d^2-a e^2 \neq 0 \land \frac{c}{a} > 0$, let $q \to \sqrt{\frac{c}{a}}$, then

$$\int \frac{1}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx \to \frac{c d+a e q}{c d^2-a e^2} \int \frac{1}{\sqrt{a+b x^2+c x^4}} \, dx - \frac{a e (e+d q)}{c d^2-a e^2} \int \frac{1+q x^2}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[1/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2]},
  (c*d+a*e*q)/(c*d^2-a*e^2)*Int[1/Sqrt[a+b*x^2+c*x^4],x] -
  (a*e*(e+d*q))/(c*d^2-a*e^2)*Int[(1+q*x^2)/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && PosQ[c/a]
```

```
Int[1/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[c/a,2]},
  (c*d+a*e*q)/(c*d^2-a*e^2)*Int[1/Sqrt[a+c*x^4],x] -
  (a*e*(e+d*q))/(c*d^2-a*e^2)*Int[(1+q*x^2)/((d+e*x^2)*Sqrt[a+c*x^4]),x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && NeQ[c*d^2-a*e^2,0] && PosQ[c/a]
```

2. $\int \frac{1}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land c d^2-a e^2 \neq 0 \land \frac{c}{a} \not> 0$

1. $\int \frac{1}{(d+e x^2)\sqrt{a+c x^4}}\,dx$ when $\frac{c}{a} \ngtr 0$

1: $\int \frac{1}{(d+e x^2)\sqrt{a+c x^4}}\,dx$ when $\frac{c}{a} \ngtr 0 \land a > 0$

Rule 1.2.2.3.7.2.1.2.2.2.1.1: If $\frac{c}{a} \ngtr 0 \land a > 0$, let $q \to \left(-\frac{c}{a}\right)^{1/4}$, then

$$\int \frac{1}{(d+e x^2)\sqrt{a+c x^4}}\,dx \longrightarrow \frac{1}{d\sqrt{a}\,q}\,\text{EllipticPi}\left[-\frac{e}{d q^2},\,\text{ArcSin}[q\,x],\,-1\right]$$

Program code:

```
Int[1/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[-c/a,4]},
  1/(d*Sqrt[a]*q)*EllipticPi[-e/(d*q^2),ArcSin[q*x],-1]] /;
FreeQ[{a,c,d,e},x] && NegQ[c/a] && GtQ[a,0]
```

2: $\int \frac{1}{(d+e x^2)\sqrt{a+c x^4}}\,dx$ when $\frac{c}{a} \ngtr 0 \land a \ngtr 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{\frac{a+c x^4}{a}}}{\sqrt{a+c x^4}} == 0$

Rule 1.2.2.3.7.1.2.2.2.1.2: If $\frac{c}{a} \ngtr 0 \land a \ngtr 0$, then

$$\int \frac{1}{(d+e x^2)\sqrt{a+c x^4}}\,dx \to \frac{\sqrt{1+\frac{c x^4}{a}}}{\sqrt{a+c x^4}} \int \frac{1}{(d+e x^2)\sqrt{1+\frac{c x^4}{a}}}\,dx$$

Program code:

```
Int[1/((d_+e_.*x_^2)*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  Sqrt[1+c*x^4/a]/Sqrt[a+c*x^4]*Int[1/((d+e*x^2)*Sqrt[1+c*x^4/a]),x] /;
FreeQ[{a,c,d,e},x] && NegQ[c/a] && Not[GtQ[a,0]]
```

2: $\int \frac{1}{\left(d+e x^2\right)\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac\neq 0 \wedge \frac{c}{a}\not>0$

Derivation: Piecewise constant extraction

Basis: Let $q\to\sqrt{b^2-4ac}$, then $\partial_x \frac{\sqrt{1+\frac{2cx^2}{b-q}}\,\sqrt{1+\frac{2cx^2}{b+q}}}{\sqrt{a+bx^2+cx^4}} == 0$

- Rule 1.2.2.3.7.1.2.2.2.2: If $b^2-4ac\neq 0 \wedge \frac{c}{a}\not>0$, let $q\to\sqrt{b^2-4ac}$, then

$$\int \frac{1}{\left(d+e x^2\right)\sqrt{a+b x^2+c x^4}}\,dx \to \frac{\sqrt{1+\frac{2cx^2}{b-q}}\,\sqrt{1+\frac{2cx^2}{b+q}}}{\sqrt{a+bx^2+cx^4}}\int \frac{1}{\left(d+e x^2\right)\sqrt{1+\frac{2cx^2}{b-q}}\,\sqrt{1+\frac{2cx^2}{b+q}}}\,dx$$

Program code:

```
Int[1/((d_+e_.*x_^2)*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  With[{q=Rt[b^2-4*a*c,2]},
  Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]/Sqrt[a+b*x^2+c*x^4]*
    Int[1/((d+e*x^2)*Sqrt[1+2*c*x^2/(b-q)]*Sqrt[1+2*c*x^2/(b+q)]),x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NegQ[c/a]
```

2: $\int \frac{(a+b x^2+c x^4)^p}{d+e x^2}\,dx$ when $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land p+\frac{1}{2}\in\mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: $\frac{1}{d+e x^2} == \frac{c d-b e-c e x^2}{c d^2-b d e+a e^2}+\frac{e^2 (a+b x^2+c x^4)}{(c d^2-b d e+a e^2)(d+e x^2)}$

Rule 1.2.2.3.7.2.2: If $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land p+\frac{1}{2}\in\mathbb{Z}^-$, then

$$\int \frac{(a+b x^2+c x^4)^p}{d+e x^2}\,dx \to \frac{1}{c d^2-b d e+a e^2}\int (c d-b e-c e x^2)(a+b x^2+c x^4)^p\,dx+\frac{e^2}{c d^2-b d e+a e^2}\int \frac{(a+b x^2+c x^4)^{p+1}}{d+e x^2}\,dx$$

Program code:

```
Int[(a_+b_.*x_^2+c_.*x_^4)^p_/(d_+e_.*x_^2),x_Symbol] :=
  1/(c*d^2-b*d*e+a*e^2)*Int[(c*d-b*e-c*e*x^2)*(a+b*x^2+c*x^4)^p,x] +
  e^2/(c*d^2-b*d*e+a*e^2)*Int[(a+b*x^2+c*x^4)^(p+1)/(d+e*x^2),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && ILtQ[p+1/2,0]
```

```
Int[(a_+c_.*x_^4)^p_/(d_+e_.*x_^2),x_Symbol] :=
  1/(c*d^2+a*e^2)*Int[(c*d-c*e*x^2)*(a+c*x^4)^p,x] +
  e^2/(c*d^2+a*e^2)*Int[(a+c*x^4)^(p+1)/(d+e*x^2),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && ILtQ[p+1/2,0]
```

2. $\int (d+e x^2)^q (a+b x^2+c x^4)^p\,dx$ when $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land p+\frac{1}{2}\in\mathbb{Z} \land q+1\in\mathbb{Z}^-$

1: $\int \frac{(d+e x^2)^q}{\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4ac\neq 0 \land q+1\in\mathbb{Z}^-$

Rule 1.2.2.3.7.2.1: If $b^2-4ac\neq 0 \land q+1\in\mathbb{Z}^-$, then

$$\int \frac{\left(d+e\,x^2\right)^q}{\sqrt{a+b\,x^2+c\,x^4}}\,dx \longrightarrow$$

$$-\frac{e^2\,x\,\left(d+e\,x^2\right)^{q+1}\,\sqrt{a+b\,x^2+c\,x^4}}{2\,d\,(q+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}+$$

$$\frac{1}{2\,d\,(q+1)\,\left(c\,d^2-b\,d\,e+a\,e^2\right)}\int \frac{1}{\sqrt{a+b\,x^2+c\,x^4}}\left(d+e\,x^2\right)^{q+1}\left(a\,e^2\,(2\,q+3)+2\,d\,\left(c\,d-b\,e\right)\,(q+1)-2\,e\,\left(c\,d\,(q+1)-b\,e\,(q+2)\right)\,x^2+c\,e^2\,(2\,q+5)\,x^4\right)dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_/Sqrt[a_+b_.*x_^2+c_.*x_^4],x_Symbol] :=
  -e^2*x*(d+e*x^2)^(q+1)*Sqrt[a+b*x^2+c*x^4]/(2*d*(q+1)*(c*d^2-b*d*e+a*e^2)) +
  1/(2*d*(q+1)*(c*d^2-b*d*e+a*e^2))*Int[(d+e*x^2)^(q+1)/Sqrt[a+b*x^2+c*x^4]*
    Simp[a*e^2*(2*q+3)+2*d*(c*d-b*e)*(q+1)-2*e*(c*d*(q+1)-b*e*(q+2))*x^2+c*e^2*(2*q+5)*x^4,x],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && ILtQ[q,-1]
```

```
Int[(d_+e_.*x_^2)^q_/Sqrt[a_+c_.*x_^4],x_Symbol] :=
  -e^2*x*(d+e*x^2)^(q+1)*Sqrt[a+c*x^4]/(2*d*(q+1)*(c*d^2+a*e^2)) +
  1/(2*d*(q+1)*(c*d^2+a*e^2))*Int[(d+e*x^2)^(q+1)/Sqrt[a+c*x^4]*
    Simp[a*e^2*(2*q+3)+2*c*d^2*(q+1)-2*e*c*d*(q+1)*x^2+c*e^2*(2*q+5)*x^4,x],x] /;
FreeQ[{a,c,d,e},x] && ILtQ[q,-1]
```

2. $\int \frac{\sqrt{a+b x^2+c x^4}}{\left(d+e x^2\right)^2} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$

1: $\int \frac{\sqrt{a+b x^2+c x^4}}{\left(d+e x^2\right)^2} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land c d^2-a e^2 == 0 \land \frac{e}{d} > 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\left(d+e x^2\right) \sqrt{\frac{e^2 \left(a+b x^2+c x^4\right)}{c \left(d+e x^2\right)^2}}}{\sqrt{a+b x^2+c x^4}} == 0$

–

Rule 1.2.2.3.7.2.2.1: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0 \land c d^2-a e^2 == 0 \land \frac{e}{d} > 0$, let $q \to \sqrt{\frac{e}{d}}$, then

$$\int \frac{\sqrt{a+b x^2+c x^4}}{\left(d+e x^2\right)^2} \, dx \to \frac{c \left(d+e x^2\right) \sqrt{\frac{e^2 \left(a+b x^2+c x^4\right)}{c \left(d+e x^2\right)^2}}}{2 d e^2 q \sqrt{a+b x^2+c x^4}} \text{EllipticE}\left[2 \text{ArcTan}[q x], \frac{2 c d-b e}{4 c d}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2+c_.*x_^4]/(d_+e_.*x_^2)^2,x_Symbol] :=
  With[{q=Rt[e/d,2]},
  c*(d+e*x^2)*Sqrt[(e^2*(a+b*x^2+c*x^4))/(c*(d+e*x^2)^2)]/(2*d*e^2*q*Sqrt[a+b*x^2+c*x^4])*
    EllipticE[2*ArcTan[q*x],(2*c*d-b*e)/(4*c*d)]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && EqQ[c*d^2-a*e^2,0] && PosQ[e/d]
```

2: $\int \frac{\sqrt{a+b x^2+c x^4}}{\left(d+e x^2\right)^2} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$

Derivation: Algebraic expansion, integration by parts and algebraic expansion

Basis: $\frac{1}{\left(d+e x^2\right)^2} = \frac{d-e x^2}{2 d \left(d+e x^2\right)^2} + \frac{1}{2 d \left(d+e x^2\right)}$

Basis: $\partial_x \frac{x}{d+e x^2} = \frac{d-e x^2}{\left(d+e x^2\right)^2}$

Basis: $\frac{a-c x^4}{d+e x^2} = \frac{c \left(d-e x^2\right)}{e^2} - \frac{c d^2-a e^2}{e^2 \left(d+e x^2\right)}$

Rule 1.2.2.3.7.2.2.2: If $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$, then

$$\int \frac{\sqrt{a+b x^2+c x^4}}{\left(d+e x^2\right)^2} \, dx \to \frac{1}{2 d} \int \frac{\left(d-e x^2\right) \sqrt{a+b x^2+c x^4}}{\left(d+e x^2\right)^2} \, dx + \frac{1}{2 d} \int \frac{\sqrt{a+b x^2+c x^4}}{d+e x^2} \, dx$$

$$\to \frac{x \sqrt{a+b x^2+c x^4}}{2 d \left(d+e x^2\right)} - \frac{1}{2 d} \int \frac{x^2 \left(b+2 c x^2\right)}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx + \frac{1}{2 d} \int \frac{\sqrt{a+b x^2+c x^4}}{d+e x^2} \, dx$$

$$\to \frac{x \sqrt{a+b x^2+c x^4}}{2 d \left(d+e x^2\right)} + \frac{1}{2 d} \int \frac{a-c x^4}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$$

$$\to \frac{x \sqrt{a+b x^2+c x^4}}{2 d \left(d+e x^2\right)} + \frac{c}{2 d e^2} \int \frac{d-e x^2}{\sqrt{a+b x^2+c x^4}} \, dx - \frac{c d^2-a e^2}{2 d e^2} \int \frac{1}{\left(d+e x^2\right) \sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2+c_.*x_^4]/(d_+e_.*x_^2)^2,x_Symbol] :=
  x*Sqrt[a+b*x^2+c*x^4]/(2*d*(d+e*x^2)) +
  c/(2*d*e^2)*Int[(d-e*x^2)/Sqrt[a+b*x^2+c*x^4],x] -
  (c*d^2-a*e^2)/(2*d*e^2)*Int[1/((d+e*x^2)*Sqrt[a+b*x^2+c*x^4]),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[Sqrt[a_+c_.*x_^4]/(d_+e_.*x_^2)^2,x_Symbol] :=
  x*Sqrt[a+c*x^4]/(2*d*(d+e*x^2)) +
  c/(2*d*e^2)*Int[(d-e*x^2)/Sqrt[a+c*x^4],x] -
  (c*d^2-a*e^2)/(2*d*e^2)*Int[1/((d+e*x^2)*Sqrt[a+c*x^4]),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0]
```

3: $\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx$ when $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land q\in\mathbb{Z}^- \land p+\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Note: Need to replace with a recurrence!

Rule 1.2.2.3.7.2.3: If $b^2-4ac\neq 0 \land c d^2-b d e+a e^2\neq 0 \land q\in\mathbb{Z}^- \land p+\frac{1}{2}\in\mathbb{Z}$, then

$$\int (d+e x^2)^q (a+b x^2+c x^4)^p \, dx \longrightarrow \int \frac{\text{ExpandIntegrand}\left[(d+e x^2)^q (a+b x^2+c x^4)^{p+\frac{1}{2}},\, x\right]}{\sqrt{a+b x^2+c x^4}} \, dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  Module[{aa,bb,cc},
  Int[ReplaceAll[ExpandIntegrand[1/Sqrt[aa+bb*x^2+cc*x^4],(d+e*x^2)^q*(aa+bb*x^2+cc*x^4)^(p+1/2),x],{aa→a,bb→b,cc→c}],x]] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0] && ILtQ[q,0] && IntegerQ[p+1/2]
```

```
Int[(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_,x_Symbol] :=
  Module[{aa,cc},
  Int[ReplaceAll[ExpandIntegrand[1/Sqrt[aa+cc*x^4],(d+e*x^2)^q*(aa+cc*x^4)^(p+1/2),x],{aa→a,cc→c}],x]] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0] && ILtQ[q,0] && IntegerQ[p+1/2]
```

8. $\int \frac{1}{\sqrt{d+e x^2}\ \sqrt{a+b x^2+c x^4}}\, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$

1. $\int \frac{1}{\sqrt{d+e x^2}\ \sqrt{a+b x^2+c x^4}}\, dx$ when $c d-b e == 0$

1: $\int \frac{1}{\sqrt{d+e x^2}\ \sqrt{a+b x^2+c x^4}}\, dx$ when $c d-b e == 0 \land a > 0 \land d > 0$

Rule 1.2.2.3.8.1.1: If $c d-b e == 0 \land a > 0 \land d > 0$, then

$$\int \frac{1}{\sqrt{d+e x^2}\ \sqrt{a+b x^2+c x^4}}\, dx \to \frac{1}{2 \sqrt{a}\ \sqrt{d}\ \sqrt{-\frac{e}{d}}} \text{EllipticF}\left[2 \text{ArcSin}\left[\sqrt{-\frac{e}{d}}\ x\right], \frac{b d}{4 a e}\right]$$

Program code:

```
Int[1/(Sqrt[d_+e_.*x_^2]*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  1/(2*Sqrt[a]*Sqrt[d]*Rt[-e/d,2])*EllipticF[2*ArcSin[Rt[-e/d,2]*x],b*d/(4*a*e)] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c*d-b*e,0] && GtQ[a,0] && GtQ[d,0]
```

2: $\int \frac{1}{\sqrt{d+e x^2}\ \sqrt{a+b x^2+c x^4}}\, dx$ when $c d-b e == 0 \land \neg (a > 0 \land d > 0)$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{\frac{a+b x^2+c x^4}{a}}\sqrt{\frac{d+e x^2}{d}}}{\sqrt{d+e x^2}\sqrt{a+b x^2+c x^4}} == 0$

Rule 1.2.2.3.8.1.2: If $c d - b e == 0 \land \neg (a > 0 \land d > 0)$, then

$$\int \frac{1}{\sqrt{d+e x^2}\sqrt{a+b x^2+c x^4}}\,dx \to \frac{\sqrt{\frac{d+e x^2}{d}}\sqrt{\frac{a+b x^2+c x^4}{a}}}{\sqrt{d+e x^2}\sqrt{a+b x^2+c x^4}}\int \frac{1}{\sqrt{1+\frac{e}{d}x^2}\sqrt{1+\frac{b}{a}x^2+\frac{c}{a}x^4}}\,dx$$

Program code:

```
Int[1/(Sqrt[d_+e_.*x_^2]*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  Sqrt[(d+e*x^2)/d]*Sqrt[(a+b*x^2+c*x^4)/a]/(Sqrt[d+e*x^2]*Sqrt[a+b*x^2+c*x^4])*
    Int[1/(Sqrt[1+e/d*x^2]*Sqrt[1+b/a*x^2+c/a*x^4]),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c*d-b*e,0] && Not[GtQ[a,0] && GtQ[d,0]]
```

2: $\int \frac{1}{\sqrt{d+e x^2}\sqrt{a+b x^2+c x^4}}\,dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x\sqrt{e+\frac{d}{x^2}}}{\sqrt{d+e x^2}} == 0$

Basis: $\partial_x \frac{x^2\sqrt{c+\frac{b}{x^2}+\frac{a}{x^4}}}{\sqrt{a+b x^2+c x^4}} == 0$

Note: The resulting integrand can be reduced to an integrand of the form $\frac{1}{\sqrt{e+d x}\sqrt{c+b x+a x^2}}$ using the substitution $x \to \frac{1}{x^2}$.

Rule 1.2.2.3.8.2: If $b^2 - 4 a c \neq 0 \land c d^2 - b d e + a e^2 \neq 0$, then

$$\int \frac{1}{\sqrt{d + e x^2} \sqrt{a + b x^2 + c x^4}} \, dx \to \frac{x^3 \sqrt{e + \frac{d}{x^2}} \sqrt{c + \frac{b}{x^2} + \frac{a}{x^4}}}{\sqrt{d + e x^2} \sqrt{a + b x^2 + c x^4}} \int \frac{1}{x^3 \sqrt{e + \frac{d}{x^2}} \sqrt{c + \frac{b}{x^2} + \frac{a}{x^4}}} \, dx$$

Program code:

```
Int[1/(Sqrt[d_+e_.*x_^2]*Sqrt[a_+b_.*x_^2+c_.*x_^4]),x_Symbol] :=
  x^3*Sqrt[e+d/x^2]*Sqrt[c+b/x^2+a/x^4]/(Sqrt[d+e*x^2]*Sqrt[a+b*x^2+c*x^4])*
    Int[1/(x^3*Sqrt[e+d/x^2]*Sqrt[c+b/x^2+a/x^4]),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[1/(Sqrt[d_+e_.*x_^2]*Sqrt[a_+c_.*x_^4]),x_Symbol] :=
  x^3*Sqrt[e+d/x^2]*Sqrt[c+a/x^4]/(Sqrt[d+e*x^2]*Sqrt[a+c*x^4])*
    Int[1/(x^3*Sqrt[e+d/x^2]*Sqrt[c+a/x^4]),x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0]
```

9. $\int \frac{\sqrt{a+b x^2+c x^4}}{\sqrt{d+e x^2}} \, dx$ when $b^2-4 a c \neq 0 \land c d^2-b d e+a e^2 \neq 0$

1. $\int \frac{\sqrt{a+b x^2+c x^4}}{\sqrt{d+e x^2}} \, dx$ when $c d-b e == 0$

1: $\int \frac{\sqrt{a+b x^2+c x^4}}{\sqrt{d+e x^2}} \, dx$ when $c d-b e == 0 \land a>0 \land d>0$

Rule 1.2.2.3.9.1.1: If $c d-b e == 0 \land a>0 \land d>0$, then

$$\int \frac{\sqrt{a+b x^2+c x^4}}{\sqrt{d+e x^2}} \, dx \to \frac{\sqrt{a}}{2 \sqrt{d} \sqrt{-\frac{e}{d}}} \text{EllipticE}\left[2 \text{ArcSin}\left[\sqrt{-\frac{e}{d}} \, x\right], \frac{b d}{4 a e}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2+c_.*x_^4]/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  Sqrt[a]/(2*Sqrt[d]*Rt[-e/d,2])*EllipticE[2*ArcSin[Rt[-e/d,2]*x],b*d/(4*a*e)] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c*d-b*e,0] && GtQ[a,0] && GtQ[d,0]
```

2: $\int \frac{\sqrt{a+b x^2+c x^4}}{\sqrt{d+e x^2}} \, dx$ when $c d-b e == 0 \land \neg (a>0 \land d>0)$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{a+b x^2+c x^4} \sqrt{\frac{d+e x^2}{d}}}{\sqrt{d+e x^2} \sqrt{\frac{a+b x^2+c x^4}{a}}} == 0$

Rule 1.2.2.3.9.1.2: If $c d-b e == 0 \land \neg (a>0 \land d>0)$, then

$$\int \frac{\sqrt{a+b x^2+c x^4}}{\sqrt{d+e x^2}}\, dx \to \frac{\sqrt{a+b x^2+c x^4}\,\sqrt{\frac{d+e x^2}{d}}}{\sqrt{d+e x^2}\,\sqrt{\frac{a+b x^2+c x^4}{a}}} \int \frac{\sqrt{1+\frac{b}{a} x^2+\frac{c}{a} x^4}}{\sqrt{1+\frac{e}{d} x^2}}\, dx$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2+c_.*x_^4]/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  Sqrt[a+b*x^2+c*x^4]*Sqrt[(d+e*x^2)/d]/(Sqrt[d+e*x^2]*Sqrt[(a+b*x^2+c*x^4)/a])*
    Int[Sqrt[1+b/a*x^2+c/a*x^4]/Sqrt[1+e/d*x^2],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c*d-b*e,0] && Not[GtQ[a,0] && GtQ[d,0]]
```

2: $\int \frac{\sqrt{a+b x^2+c x^4}}{\sqrt{d+e x^2}}\, dx$ when $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x \sqrt{e+\frac{d}{x^2}}}{\sqrt{d+e x^2}} = 0$

Basis: $\partial_x \frac{\sqrt{a+b x^2+c x^4}}{x^2 \sqrt{c+\frac{b}{x^2}+\frac{a}{x^4}}} = 0$

Note: The resulting integrand can be reduced to an integrand of the form $\frac{1}{\sqrt{e+d x}\,\sqrt{c+b x+a x^2}}$ using the substitution $x \to \frac{1}{x^2}$.

Rule 1.2.2.3.9.2: If $b^2-4 a c \neq 0 \wedge c d^2-b d e+a e^2 \neq 0$, then

$$\int \frac{\sqrt{a+b x^2+c x^4}}{\sqrt{d+e x^2}}\,dx \longrightarrow \frac{\sqrt{e+\frac{d}{x^2}}\;\sqrt{a+b x^2+c x^4}}{x\sqrt{d+e x^2}\;\sqrt{c+\frac{b}{x^2}+\frac{a}{x^4}}}\int \frac{x\sqrt{c+\frac{b}{x^2}+\frac{a}{x^4}}}{\sqrt{e+\frac{d}{x^2}}}\,dx$$

Program code:

```
Int[Sqrt[a_+b_.*x_^2+c_.*x_^4]/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  Sqrt[e+d/x^2]*Sqrt[a+b*x^2+c*x^4]/(x*Sqrt[d+e*x^2]*Sqrt[c+b/x^2+a/x^4])*
    Int[(x*Sqrt[c+b/x^2+a/x^4])/Sqrt[e+d/x^2],x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[b^2-4*a*c,0] && NeQ[c*d^2-b*d*e+a*e^2,0]
```

```
Int[Sqrt[a_+c_.*x_^4]/Sqrt[d_+e_.*x_^2],x_Symbol] :=
  Sqrt[e+d/x^2]*Sqrt[a+c*x^4]/(x*Sqrt[d+e*x^2]*Sqrt[c+a/x^4])*
    Int[(x*Sqrt[c+a/x^4])/Sqrt[e+d/x^2],x] /;
FreeQ[{a,c,d,e},x] && NeQ[c*d^2+a*e^2,0]
```

10: $\int (d+e x^2)^q (a+b x^2+c x^4)^p\,dx$ when $b^2-4ac\neq 0 \land ((p \mid q)\in\mathbb{Z} \lor p\in\mathbb{Z}^+ \lor q\in\mathbb{Z}^+)$

Derivation: Algebraic expansion

Rule 1.2.2.3.10: If $b^2-4ac\neq 0 \land ((p \mid q)\in\mathbb{Z} \lor p\in\mathbb{Z}^+ \lor q\in\mathbb{Z}^+)$, then

$$\int (d+e x^2)^q (a+b x^2+c x^4)^p\,dx \longrightarrow \int \text{ExpandIntegrand}\left[(d+e x^2)^q (a+b x^2+c x^4)^p,\ x\right]dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_+b_.*x_^2+c_.*x_^4)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^q*(a+b*x^2+c*x^4)^p,x],x] /;
FreeQ[{a,b,c,d,e,p,q},x] && NeQ[b^2-4*a*c,0] && (IntegerQ[p] && IntegerQ[q] || IGtQ[p,0] || IGtQ[q,0])
```

```
Int[(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(d+e*x^2)^q*(a+c*x^4)^p,x],x] /;
FreeQ[{a,c,d,e,p,q},x] && (IntegerQ[p] && IntegerQ[q] || IGtQ[p,0])
```

11: $\int (d+e x^2)^q (a+c x^4)^p \, dx$ when $c d^2 + a e^2 \neq 0 \land p \notin \mathbb{Z} \land q \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: If $q \in \mathbb{Z}$, then $(d+e x^2)^q == \left(\frac{d}{d^2-e^2 x^4} - \frac{e x^2}{d^2-e^2 x^4}\right)^{-q}$

Note: Resulting integrands are of the form $x^m (a+b x^4)^p (c+d x^4)^q$ which are integrable in terms of the Appell hypergeometric function .

Rule 1.2.2.3.11: If $c d^2 + a e^2 \neq 0 \land p \notin \mathbb{Z} \land q \in \mathbb{Z}^-$, then

$$\int (d+e x^2)^q (a+c x^4)^p \, dx \to \int (a+c x^4)^p \, \text{ExpandIntegrand}\left[\left(\frac{d}{d^2-e^2 x^4} - \frac{e x^2}{d^2-e^2 x^4}\right)^{-q}, x\right] dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_*(a_+c_.*x_^4)^p_,x_Symbol] :=
  Int[ExpandIntegrand[(a+c*x^4)^p,(d/(d^2-e^2*x^4)-e*x^2/(d^2-e^2*x^4))^(-q),x],x] /;
FreeQ[{a,c,d,e,p},x] && NeQ[c*d^2+a*e^2,0] && Not[IntegerQ[p]] && ILtQ[q,0]
```

U: $\int \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p\,dx$

Rule 1.2.2.3.U:

$$\int \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p\,dx \longrightarrow \int \left(d+e\,x^2\right)^q \left(a+b\,x^2+c\,x^4\right)^p\,dx$$

Program code:

```
Int[(d_+e_.*x_^2)^q_.*(a_+b_.*x_^2+c_.*x_^4)^p_.,x_Symbol] :=
  Unintegrable[(d+e*x^2)^q*(a+b*x^2+c*x^4)^p,x] /;
FreeQ[{a,b,c,d,e,p,q},x]
```

```
Int[(d_+e_.*x_^2)^q_.*(a_+c_.*x_^4)^p_.,x_Symbol] :=
  Unintegrable[(d+e*x^2)^q*(a+c*x^4)^p,x] /;
FreeQ[{a,c,d,e,p,q},x]
```