Rules for integrands of the form $\text{Trig}[d+e\,x]^m\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p$

1. $\int\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$

  1. $\int\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0$

    1: $\int\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0\land p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2-4\,a\,c==0$, then $a+b\,z+c\,z^2==\frac{(b+2\,c\,z)^2}{4\,c}$

Rule: If $b^2-4\,a\,c==0\land p\in\mathbb{Z}$, then

$$\int\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx\;\longrightarrow\;\frac{1}{4^p\,c^p}\int\left(b+2\,c\,\text{Sin}[d+e\,x]^n\right)^{2\,p}\,dx$$

Program code:

```
Int[(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[(b+2*c*Sin[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

```
Int[(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[(b+2*c*Cos[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

    2: $\int\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0\land p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c==0$, then $\partial_x\frac{\left(a+b\,F[x]+c\,F[x]^2\right)^p}{(b+2\,c\,F[x])^{2\,p}}==0$

Rule: If $b^2-4\,a\,c==0\land p\notin\mathbb{Z}$, then

$$\int \left(a + b \operatorname{Sin}[d + e x]^n + c \operatorname{Sin}[d + e x]^{2n}\right)^p dx \longrightarrow \frac{\left(a + b \operatorname{Sin}[d + e x]^n + c \operatorname{Sin}[d + e x]^{2n}\right)^p}{\left(b + 2 c \operatorname{Sin}[d + e x]^n\right)^{2p}} \int \left(b + 2 c \operatorname{Sin}[d + e x]^n\right)^{2p} dx$$

Program code:

```
Int[(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Sin[d+e*x]^n+c*Sin[d+e*x]^(2*n))^p/(b+2*c*Sin[d+e*x]^n)^(2*p)*Int[u*(b+2*c*Sin[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

```
Int[(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Cos[d+e*x]^n+c*Cos[d+e*x]^(2*n))^p/(b+2*c*Cos[d+e*x]^n)^(2*p)*Int[u*(b+2*c*Cos[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2. $\int (a + b \operatorname{Sin}[d + e x]^n + c \operatorname{Sin}[d + e x]^{2n})^p \, dx$ when $b^2 - 4 a c \neq 0$

1: $\int \frac{1}{a + b \operatorname{Sin}[d + e x]^n + c \operatorname{Sin}[d + e x]^{2n}} \, dx$ when $b^2 - 4 a c \neq 0$

Derivation: Algebraic expansion

Basis: If $q = \sqrt{b^2 - 4 a c}$, then $\frac{1}{a+b z+c z^2} == \frac{2 c}{q (b-q+2 c z)} - \frac{2 c}{q (b+q+2 c z)}$

Rule: If $b^2 - 4 a c \neq 0$, let $q = \sqrt{b^2 - 4 a c}$, then

$$\int \frac{1}{a + b \operatorname{Sin}[d + e x]^n + c \operatorname{Sin}[d + e x]^{2n}} \, dx \to \frac{2 c}{q} \int \frac{1}{b - q + 2 c \operatorname{Sin}[d + e x]^n} \, dx - \frac{2 c}{q} \int \frac{1}{b + q + 2 c \operatorname{Sin}[d + e x]^n} \, dx$$

Program code:

```
Int[1/(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[1/(b-q+2*c*Sin[d+e*x]^n),x] -
  2*c/q*Int[1/(b+q+2*c*Sin[d+e*x]^n),x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0]
```

```
Int[1/(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  2*c/q*Int[1/(b-q+2*c*Cos[d+e*x]^n),x] -
  2*c/q*Int[1/(b+q+2*c*Cos[d+e*x]^n),x]] /;
FreeQ[{a,b,c,d,e,n},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0]
```

2. $\int \text{Sin}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$

   1. $\int \text{Sin}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0$

      1: $\int \text{Sin}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2-4\,a\,c==0$, then $a+b\,z+c\,z^2==\frac{(b+2\,c\,z)^2}{4\,c}$

Rule: If $b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$, then

$$\int \text{Sin}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx\;\rightarrow\;\frac{1}{4^p\,c^p}\int \text{Sin}[d+e\,x]^m \left(b+2\,c\,\text{Sin}[d+e\,x]^n\right)^{2\,p}\,dx$$

Program code:

```
Int[sin[d_.+e_.*x_]^m_.*(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  1/(4^p*c^p)*Int[Sin[d+e*x]^m*(b+2*c*Sin[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

```
Int[cos[d_.+e_.*x_]^m_.*(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  1/(4^p*c^p)*Int[Cos[d+e*x]^m*(b+2*c*Cos[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

2: $\int \text{Sin}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p \, dx$ when $b^2-4\,a\,c==0 \land p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c==0$, then $\partial_x \frac{\left(a+b\,F[x]+c\,F[x]^2\right)^p}{\left(b+2\,c\,F[x]\right)^{2\,p}}==0$

Rule: If $b^2-4\,a\,c==0 \land p\notin\mathbb{Z}$, then

$$\int \text{Sin}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx \to \frac{\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p}{\left(b+2\,c\,\text{Sin}[d+e\,x]^n\right)^{2\,p}} \int \text{Sin}[d+e\,x]^m \left(b+2\,c\,\text{Sin}[d+e\,x]^n\right)^{2\,p} dx$$

Program code:

```
Int[sin[d_.+e_.*x_]^m_.*(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Sin[d+e*x]^n+c*Sin[d+e*x]^(2*n))^p/(b+2*c*Sin[d+e*x]^n)^(2*p)*Int[Sin[d+e*x]^m*(b+2*c*Sin[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

```
Int[cos[d_.+e_.*x_]^m_.*(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Cos[d+e*x]^n+c*Cos[d+e*x]^(2*n))^p/(b+2*c*Cos[d+e*x]^n)^(2*p)*Int[Cos[d+e*x]^m*(b+2*c*Cos[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2. $\int \text{Sin}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p \, dx$ when $b^2-4\,a\,c\neq 0$

1: $\int \text{Sin}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p \, dx$ when $\frac{m}{2}\in\mathbb{Z} \land b^2-4\,a\,c\neq 0 \land \frac{n}{2}\in\mathbb{Z} \land p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Sin}[z]^2 == \frac{1}{1+\text{Cot}[z]^2}$

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then

$$\text{Sin}[d+e\,x]^m\,F\left[\text{Sin}[d+e\,x]^2\right]=-\frac{1}{e}\,\text{Subst}\left[\frac{F\left[\frac{1}{1+x^2}\right]}{\left(1+x^2\right)^{m/2+1}},\,x,\,\text{Cot}[d+e\,x]\right]\partial_x\text{Cot}[d+e\,x]$$

Rule: If $\frac{m}{2}\in\mathbb{Z}\,\land\,b^2-4\,a\,c\neq 0\,\land\,\frac{n}{2}\in\mathbb{Z}\,\land\,p\in\mathbb{Z}$, then

$$\int\text{Sin}[d+e\,x]^m\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx\;\rightarrow\;-\frac{1}{e}\,\text{Subst}\left[\int\frac{\left(c+b\left(1+x^2\right)^{n/2}+a\left(1+x^2\right)^n\right)^p}{\left(1+x^2\right)^{m/2+n\,p+1}}\,dx,\,x,\,\text{Cot}[d+e\,x]\right]$$

Program code:

```
Int[sin[d_.+e_.*x_]^m_*(a_.+b_.*sin[d_.+e_.*x_]^n_+c_.*sin[d_.+e_.*x_]^n2_)^p_,x_Symbol] :=
  Module[{f=FreeFactors[Cot[d+e*x],x]},
  -f/e*Subst[Int[ExpandToSum[c+b*(1+x^2)^(n/2)+a*(1+x^2)^n,x]^p/(1+f^2*x^2)^(m/2+n*p+1),x],x,Cot[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[m/2] && NeQ[b^2-4*a*c,0] && IntegerQ[n/2] && IntegerQ[p]
```

```
Int[cos[d_.+e_.*x_]^m_.*(a_.+b_.*cos[d_.+e_.*x_]^n_+c_.*cos[d_.+e_.*x_]^n2_)^p_,x_Symbol] :=
  Module[{f=FreeFactors[Tan[d+e*x],x]},
  f/e*Subst[Int[ExpandToSum[c+b*(1+x^2)^(n/2)+a*(1+x^2)^n,x]^p/(1+f^2*x^2)^(m/2+n*p+1),x],x,Tan[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[m/2] && NeQ[b^2-4*a*c,0] && IntegerQ[n/2] && IntegerQ[p]
```

2: $\int\text{Sin}[d+e\,x]^m\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $b^2-4\,a\,c\neq 0\,\land\,(m\mid n\mid p)\in\mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $b^2-4\,a\,c\neq 0\,\land\,(m\mid n\mid p)\in\mathbb{Z}$, then

$$\int\text{Sin}[d+e\,x]^m\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx\;\rightarrow\;\int\text{ExpandTrig}\left[\text{Sin}[d+e\,x]^m\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p,\,x\right]dx$$

Proeram code:

```
Int[sin[d_.+e_.*x_]^m_.*(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  Int[ExpandTrig[sin[d+e*x]^m*(a+b*sin[d+e*x]^n+c*sin[d+e*x]^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IntegersQ[m,n,p]
```

```
Int[cos[d_.+e_.*x_]^m_.*(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  Int[ExpandTrig[cos[d+e*x]^m*(a+b*cos[d+e*x]^n+c*cos[d+e*x]^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && NeQ[b^2-4*a*c,0] && IntegersQ[m,n,p]
```

3. $\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx$

1: $\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m-1}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, then

$$\text{Cos}[d+e\,x]^m\,F[\text{Sin}[d+e\,x]] = \frac{1}{e}\,\text{Subst}\left[\left(1-x^2\right)^{\frac{m-1}{2}}F[x],\,x,\,\text{Sin}[d+e\,x]\right]\partial_x\text{Sin}[d+e\,x]$$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}$, then

$$\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx \to \frac{1}{e}\,\text{Subst}\left[\int \left(1-x^2\right)^{\frac{m-1}{2}}\left(a+b\,x^n+c\,x^{2\,n}\right)^p dx,\,x,\,\text{Sin}[d+e\,x]\right]$$

Program code:

```
Int[cos[d_.+e_.*x_]^m_.*(a_.+b_.*(f_.*sin[d_.+e_.*x_])^n_.+c_.*(f_.*sin[d_.+e_.*x_])^n2_.)^p_.,x_Symbol] :=
  Module[{g=FreeFactors[Sin[d+e*x],x]},
  g/e*Subst[Int[(1-g^2*x^2)^((m-1)/2)*(a+b*(f*g*x)^n+c*(f*g*x)^(2*n))^p,x],x,Sin[d+e*x]/g]] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && EqQ[n2,2*n] && IntegerQ[(m-1)/2]
```

```
Int[sin[d_.+e_.*x_]^m_.*(a_.+b_.*(f_.*cos[d_.+e_.*x_])^n_.+c_.*(f_.*cos[d_.+e_.*x_])^n2_.)^p_.,x_Symbol] :=
  Module[{g=FreeFactors[Cos[d+e*x],x]},
  -g/e*Subst[Int[(1-g^2*x^2)^((m-1)/2)*(a+b*(f*g*x)^n+c*(f*g*x)^(2*n))^p,x],x,Cos[d+e*x]/g]] /;
FreeQ[{a,b,c,d,e,f,n,p},x] && EqQ[n2,2*n] && IntegerQ[(m-1)/2]
```

2. $\int \text{Cos}[d+e\,x]^m\,\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $\frac{m-1}{2}\notin\mathbb{Z}$

   1. $\int \text{Cos}[d+e\,x]^m\,\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c==0$

      1: $\int \text{Cos}[d+e\,x]^m\,\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2-4\,a\,c==0$, then $a+b\,z+c\,z^2==\frac{(b+2\,c\,z)^2}{4\,c}$

Rule: If $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$, then

$$\int \text{Cos}[d+e\,x]^m\,\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx\;\longrightarrow\;\frac{1}{4^p\,c^p}\int \text{Cos}[d+e\,x]^m\,\left(b+2\,c\,\text{Sin}[d+e\,x]^n\right)^{2\,p}\,dx$$

Program code:

```
Int[cos[d_.+e_.*x_]^m_*(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[Cos[d+e*x]^m*(b+2*c*Sin[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

```
Int[sin[d_.+e_.*x_]^m_*(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[Sin[d+e*x]^m*(b+2*c*Cos[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

2: $\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p \,dx$ when $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c==0\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c==0$, then $\partial_x \frac{\left(a+b\,F[x]+c\,F[x]^2\right)^p}{(b+2\,c\,F[x])^{2\,p}}==0$

Rule: If $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c==0\,\land\,p\notin\mathbb{Z}$, then

$$\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx \to \frac{\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p}{\left(b+2\,c\,\text{Sin}[d+e\,x]^n\right)^{2\,p}}\int \text{Cos}[d+e\,x]^m \left(b+2\,c\,\text{Sin}[d+e\,x]^n\right)^{2\,p} dx$$

Program code:

```
Int[cos[d_.+e_.*x_]^m_*(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Sin[d+e*x]^n+c*Sin[d+e*x]^(2*n))^p/(b+2*c*Sin[d+e*x]^n)^(2*p)*Int[Cos[d+e*x]^m*(b+2*c*Sin[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

```
Int[sin[d_.+e_.*x_]^m_*(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Cos[d+e*x]^n+c*Cos[d+e*x]^(2*n))^p/(b+2*c*Cos[d+e*x]^n)^(2*p)*Int[Sin[d+e*x]^m*(b+2*c*Cos[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2. $\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c\neq 0$

1: $\int \text{Cos}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m}{2}\in\mathbb{Z}\,\land\,b^2-4\,a\,c\neq 0\,\land\,\frac{n}{2}\in\mathbb{Z}\,\land\,p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Sin}[z]^2==\frac{1}{1+\text{Cot}[z]^2}$

Basis: $\text{Cos}[z]^2==\frac{\text{Cot}[z]^2}{1+\text{Cot}[z]^2}$

Basis: If $\frac{m}{2} \in \mathbb{Z}$, then

$$\text{Cos}[d+e\,x]^m\,F\left[\text{Sin}[d+e\,x]^2\right] = -\frac{1}{e}\,\text{Subst}\left[\frac{x^m\,F\left[\frac{1}{1+x^2}\right]}{\left(1+x^2\right)^{m/2+1}},\,x,\,\text{Cot}[d+e\,x]\right]\,\partial_x\,\text{Cot}[d+e\,x]$$

Rule: If $\frac{m}{2} \in \mathbb{Z} \wedge b^2-4\,a\,c \neq 0 \wedge \frac{n}{2} \in \mathbb{Z} \wedge p \in \mathbb{Z}$, then

$$\int \text{Cos}[d+e\,x]^m\,\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx \to -\frac{1}{e}\,\text{Subst}\left[\int \frac{x^m\,\left(c+b\,\left(1+x^2\right)^{n/2}+a\,\left(1+x^2\right)^n\right)^p}{\left(1+x^2\right)^{m/2+n\,p+1}}\,dx,\,x,\,\text{Cot}[d+e\,x]\right]$$

Program code:

```
Int[cos[d_.+e_.*x_]^m_*(a_.+b_.*sin[d_.+e_.*x_]^n_+c_.*sin[d_.+e_.*x_]^n2_)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Cot[d+e*x],x]},
  -f^(m+1)/e*Subst[Int[x^m*ExpandToSum[c+b*(1+x^2)^(n/2)+a*(1+x^2)^n,x]^p/(1+f^2*x^2)^(m/2+n*p+1),x],x,Cot[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[m/2] && NeQ[b^2-4*a*c,0] && IntegerQ[n/2] && IntegerQ[p]
```

```
Int[sin[d_.+e_.*x_]^m_.*(a_.+b_.*cos[d_.+e_.*x_]^n_+c_.*cos[d_.+e_.*x_]^n2_)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Tan[d+e*x],x]},
  f^(m+1)/e*Subst[Int[x^m*ExpandToSum[c+b*(1+x^2)^(n/2)+a*(1+x^2)^n,x]^p/(1+f^2*x^2)^(m/2+n*p+1),x],x,Tan[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[m/2] && NeQ[b^2-4*a*c,0] && IntegerQ[n/2] && IntegerQ[p]
```

2: $\int \mathrm{Cos}[d+e\,x]^m \left(a+b\,\mathrm{Sin}[d+e\,x]^n+c\,\mathrm{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m}{2}\in\mathbb{Z}\,\land\,b^2-4\,a\,c\neq 0\,\land\,(n\mid p)\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\mathrm{Cos}[z]^2 == 1-\mathrm{Sin}[z]^2$

Rule: If $\frac{m}{2}\in\mathbb{Z}\,\land\,b^2-4\,a\,c\neq 0\,\land\,(n\mid p)\in\mathbb{Z}$, then

$$\int \mathrm{Cos}[d+e\,x]^m \left(a+b\,\mathrm{Sin}[d+e\,x]^n+c\,\mathrm{Sin}[d+e\,x]^{2\,n}\right)^p dx \to \int \mathrm{ExpandTrig}\left[\left(1-\mathrm{Sin}[d+e\,x]^2\right)^{m/2}\left(a+b\,\mathrm{Sin}[d+e\,x]^n+c\,\mathrm{Sin}[d+e\,x]^{2\,n}\right)^p,\,x\right] dx$$

Proeram code:

```
Int[cos[d_.+e_.*x_]^m_.*(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Int[ExpandTrig[(1-sin[d+e*x]^2)^(m/2)*(a+b*sin[d+e*x]^n+c*sin[d+e*x]^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[m/2] && NeQ[b^2-4*a*c,0] && IntegersQ[n,p]
```

```
Int[sin[d_.+e_.*x_]^m_.*(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Int[ExpandTrig[(1-cos[d+e*x]^2)^(m/2)*(a+b*cos[d+e*x]^n+c*cos[d+e*x]^(2*n))^p,x],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[m/2] && NeQ[b^2-4*a*c,0] && IntegersQ[n,p]
```

4. $\int \mathrm{Tan}[d+e\,x]^m \left(a+b\,\mathrm{Sin}[d+e\,x]^n+c\,\mathrm{Sin}[d+e\,x]^{2\,n}\right)^p dx$

1: $\int \mathrm{Tan}[d+e\,x]^m \left(a+b\,\mathrm{Sin}[d+e\,x]^n+c\,\mathrm{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m-1}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\mathrm{Tan}[z]^2 == \frac{\mathrm{Sin}[z]^2}{1-\mathrm{Sin}[z]^2}$

Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, then

$$\mathrm{Tan}[d+e\,x]^m\,F[\mathrm{Sin}[d+e\,x]] == \frac{1}{e}\,\mathrm{Subst}\left[\frac{x^m\,F[x]}{\left(1-x^2\right)^{\frac{m+1}{2}}},\,x,\,\mathrm{Sin}[d+e\,x]\right]\partial_x\,\mathrm{Sin}[d+e\,x]$$

Rule: If $\frac{m-1}{2} \in \mathbb{Z} \land 2\,p \in \mathbb{Z}$, then

$$\int \mathrm{Tan}[d+e\,x]^m \left(a+b\,\mathrm{Sin}[d+e\,x]^n+c\,\mathrm{Sin}[d+e\,x]^{2\,n}\right)^p dx \longrightarrow \frac{1}{e}\,\mathrm{Subst}\left[\int \frac{x^m \left(a+b\,x^n+c\,x^{2\,n}\right)^p}{\left(1-x^2\right)^{\frac{m+1}{2}}} dx,\,x,\,\mathrm{Sin}[d+e\,x]\right]$$

Program code:

```
Int[tan[d_.+e_.*x_]^m_.*(a_+b_.*(f_.*sin[d_.+e_.*x_])^n_+c_.*(f_.*sin[d_.+e_.*x_])^n2_.)^p_.,x_Symbol] :=
  Module[{g=FreeFactors[Sin[d+e*x],x]},
  g^(m+1)/e*Subst[Int[x^m*(a+b*(f*g*x)^n+c*(f*g*x)^(2*n))^p/(1-g^2*x^2)^((m+1)/2),x],x,Sin[d+e*x]/g]] /;
FreeQ[{a,b,c,d,e,f,n},x] && IntegerQ[(m-1)/2] && IntegerQ[2*p]
```

```
Int[cot[d_.+e_.*x_]^m_.*(a_+b_.*(f_.*cos[d_.+e_.*x_])^n_+c_.*(f_.*cos[d_.+e_.*x_])^n2_.)^p_.,x_Symbol] :=
  Module[{g=FreeFactors[Cos[d+e*x],x]},
  -g^(m+1)/e*Subst[Int[x^m*(a+b*(f*g*x)^n+c*(f*g*x)^(2*n))^p/(1-g^2*x^2)^((m+1)/2),x],x,Cos[d+e*x]/g]] /;
FreeQ[{a,b,c,d,e,f,n},x] && IntegerQ[(m-1)/2] && IntegerQ[2*p]
```

2. $\int \text{Tan}[d+e\,x]^m\,\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $\frac{m-1}{2}\notin\mathbb{Z}$

   1. $\int \text{Tan}[d+e\,x]^m\,\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c==0$

      1: $\int \text{Tan}[d+e\,x]^m\,\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2-4\,a\,c==0$, then $a+b\,z+c\,z^2==\frac{(b+2\,c\,z)^2}{4\,c}$

Rule: If $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$, then

$$\int \text{Tan}[d+e\,x]^m\,\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx\;\longrightarrow\;\frac{1}{4^p\,c^p}\int \text{Tan}[d+e\,x]^m\,\left(b+2\,c\,\text{Sin}[d+e\,x]^n\right)^{2\,p}\,dx$$

Program code:

```
Int[tan[d_.+e_.*x_]^m_*(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[Tan[d+e*x]^m*(b+2*c*Sin[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

```
Int[cot[d_.+e_.*x_]^m_*(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[Cot[d+e*x]^m*(b+2*c*Cos[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

2: $\int \operatorname{Tan}[d+e\,x]^m \left(a+b\operatorname{Sin}[d+e\,x]^n+c\operatorname{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m-1}{2}\notin\mathbb{Z}\wedge b^2-4\,a\,c==0\wedge p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c==0$, then $\partial_x \frac{\left(a+b\,F[x]+c\,F[x]^2\right)^p}{(b+2\,c\,F[x])^{2\,p}}==0$

Rule: If $\frac{m-1}{2}\notin\mathbb{Z}\wedge b^2-4\,a\,c==0\wedge p\notin\mathbb{Z}$, then

$$\int \operatorname{Tan}[d+e\,x]^m \left(a+b\operatorname{Sin}[d+e\,x]^n+c\operatorname{Sin}[d+e\,x]^{2\,n}\right)^p dx \to \frac{\left(a+b\operatorname{Sin}[d+e\,x]^n+c\operatorname{Sin}[d+e\,x]^{2\,n}\right)^p}{\left(b+2\,c\operatorname{Sin}[d+e\,x]^n\right)^{2\,p}}\int \operatorname{Tan}[d+e\,x]^m \left(b+2\,c\operatorname{Sin}[d+e\,x]^n\right)^{2\,p} dx$$

Program code:

```
Int[tan[d_.+e_.*x_]^m_*(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Sin[d+e*x]^n+c*Sin[d+e*x]^(2*n))^p/(b+2*c*Sin[d+e*x]^n)^(2*p)*Int[Tan[d+e*x]^m*(b+2*c*Sin[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

```
Int[cot[d_.+e_.*x_]^m_*(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Cos[d+e*x]^n+c*Cos[d+e*x]^(2*n))^p/(b+2*c*Cos[d+e*x]^n)^(2*p)*Int[Cot[d+e*x]^m*(b+2*c*Cos[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2. $\int \operatorname{Tan}[d+e\,x]^m\,\left(a+b\operatorname{Sin}[d+e\,x]^n+c\operatorname{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c\neq0$

1: $\int \operatorname{Tan}[d+e\,x]^m\,\left(a+b\operatorname{Sin}[d+e\,x]^n+c\operatorname{Sin}[d+e\,x]^{2\,n}\right)^p\,dx$ when $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c\neq0\,\land\,\frac{n}{2}\in\mathbb{Z}\,\land\,p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\operatorname{Sin}[z]^2 == \frac{\operatorname{Tan}[z]^2}{1+\operatorname{Tan}[z]^2}$

Basis: $\operatorname{Tan}[d+e\,x]^m\,F\left[\operatorname{Sin}[d+e\,x]^2\right] == \frac{1}{e}\operatorname{Subst}\left[\frac{x^m\,F\left[\frac{x^2}{1+x^2}\right]}{1+x^2},\,x,\,\operatorname{Tan}[d+e\,x]\right]\,\partial_x\operatorname{Tan}[d+e\,x]$

Rule: If $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c\neq0\,\land\,\frac{n}{2}\in\mathbb{Z}\,\land\,p\in\mathbb{Z}$, then

$$\int \operatorname{Tan}[d+e\,x]^m\,\left(a+b\operatorname{Sin}[d+e\,x]^n+c\operatorname{Sin}[d+e\,x]^{2\,n}\right)^p\,dx \rightarrow \frac{1}{e}\operatorname{Subst}\left[\int\frac{x^m\,\left(c\,x^{2\,n}+b\,x^n\,\left(1+x^2\right)^{n/2}+a\,\left(1+x^2\right)^n\right)^p}{\left(1+x^2\right)^{n\,p+1}}\,dx,\,x,\,\operatorname{Tan}[d+e\,x]\right]$$

Program code:

```
Int[tan[d_.+e_.*x_]^m_.*(a_.+b_.*sin[d_.+e_.*x_]^n_+c_.*sin[d_.+e_.*x_]^n2_)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Tan[d+e*x],x]},
  f^(m+1)/e*Subst[Int[x^m*ExpandToSum[c*x^(2*n)+b*x^n*(1+x^2)^(n/2)+a*(1+x^2)^n,x]^p/(1+f^2*x^2)^(n*p+1),x],x,Tan[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e,m},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && NeQ[b^2-4*a*c,0] && IntegerQ[n/2] && IntegerQ[p]
```

```
Int[cot[d_.+e_.*x_]^m_.*(a_.+b_.*cos[d_.+e_.*x_]^n_+c_.*cos[d_.+e_.*x_]^n2_)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Cot[d+e*x],x]},
  -f^(m+1)/e*Subst[Int[x^m*ExpandToSum[c*x^(2*n)+b*x^n*(1+x^2)^(n/2)+a*(1+x^2)^n,x]^p/(1+f^2*x^2)^(n*p+1),x],x,Cot[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e,m},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && NeQ[b^2-4*a*c,0] && IntegerQ[n/2] && IntegerQ[p]
```

2: $\int \text{Tan}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m}{2}\in\mathbb{Z}\land b^2-4\,a\,c\neq 0\land(n\mid p)\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\text{Tan}[z]^2 == \frac{\text{Sin}[z]^2}{1-\text{Sin}[z]^2}$

Rule: If $\frac{m}{2}\in\mathbb{Z}\land b^2-4\,a\,c\neq 0\land(n\mid p)\in\mathbb{Z}$, then

$$\int \text{Tan}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx \to \int \text{ExpandTrig}\left[\frac{\text{Sin}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p}{\left(1-\text{Sin}[d+e\,x]^2\right)^{m/2}},\,x\right] dx$$

Proeram code:

```
Int[tan[d_.+e_.*x_]^m_.*(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Int[ExpandTrig[sin[d+e*x]^m*(a+b*sin[d+e*x]^n+c*sin[d+e*x]^(2*n))^p/(1-sin[d+e*x]^2)^(m/2),x],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[m/2] && NeQ[b^2-4*a*c,0] && IntegersQ[n,p]
```

```
Int[cot[d_.+e_.*x_]^m_.*(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Int[ExpandTrig[cos[d+e*x]^m*(a+b*cos[d+e*x]^n+c*cos[d+e*x]^(2*n))^p/(1-cos[d+e*x]^2)^(m/2),x],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[m/2] && NeQ[b^2-4*a*c,0] && IntegersQ[n,p]
```

5. $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx$

1: $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m-1}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Cot}[z]^2 == \frac{1-\text{Sin}[z]^2}{\text{Sin}[z]^2}$

Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, then

$$\text{Cot}[d+e\,x]^m\,F[\text{Sin}[d+e\,x]]==\frac{1}{e}\,\text{Subst}\left[\frac{\left(1-x^2\right)^{\frac{m-1}{2}}F[x]}{x^m},\,x,\,\text{Sin}[d+e\,x]\right]\partial_x\text{Sin}[d+e\,x]$$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}\,\wedge\,2\,p\in\mathbb{Z}$, then

$$\int\text{Cot}[d+e\,x]^m\left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p\,dx\,\longrightarrow\,\frac{1}{e}\,\text{Subst}\left[\int\frac{\left(1-x^2\right)^{\frac{m-1}{2}}\left(a+b\,x^n+c\,x^{2\,n}\right)^p}{x^m}\,dx,\,x,\,\text{Sin}[d+e\,x]\right]$$

Program code:

```
Int[cot[d_.+e_.*x_]^m_.*(a_+b_.*(f_.*sin[d_.+e_.*x_])^n_+c_.*(f_.*sin[d_.+e_.*x_])^n2_.)^p_.,x_Symbol] :=
  Module[{g=FreeFactors[Sin[d+e*x],x]},
  g^(m+1)/e*Subst[Int[(1-g^2*x^2)^((m-1)/2)*(a+b*(f*g*x)^n+c*(f*g*x)^(2*n))^p/x^m,x],x,Sin[d+e*x]/g]] /;
FreeQ[{a,b,c,d,e,f,n},x] && IntegerQ[(m-1)/2] && IntegerQ[2*p]
```

```
Int[tan[d_.+e_.*x_]^m_.*(a_+b_.*(f_.*cos[d_.+e_.*x_])^n_+c_.*(f_.*cos[d_.+e_.*x_])^n2_.)^p_.,x_Symbol] :=
  Module[{g=FreeFactors[Cos[d+e*x],x]},
  -g^(m+1)/e*Subst[Int[(1-g^2*x^2)^((m-1)/2)*(a+b*(f*g*x)^n+c*(f*g*x)^(2*n))^p/x^m,x],x,Cos[d+e*x]/g]] /;
FreeQ[{a,b,c,d,e,f,n},x] && IntegerQ[(m-1)/2] && IntegerQ[2*p]
```

2. $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m-1}{2} \notin \mathbb{Z}$

   1. $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m-1}{2} \notin \mathbb{Z} \land b^2-4\,a\,c == 0$

      1: $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m-1}{2} \notin \mathbb{Z} \land b^2-4\,a\,c == 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2-4\,a\,c == 0$, then $a+b\,z+c\,z^2 == \frac{(b+2\,c\,z)^2}{4\,c}$

Rule: If $\frac{m-1}{2} \notin \mathbb{Z} \land b^2-4\,a\,c == 0 \land p \in \mathbb{Z}$, then

$$\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx \rightarrow \frac{1}{4^p\,c^p}\int \text{Cot}[d+e\,x]^m \left(b+2\,c\,\text{Sin}[d+e\,x]^n\right)^{2\,p} dx$$

Program code:

```
Int[cot[d_.+e_.*x_]^m_*(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[Cot[d+e*x]^m*(b+2*c*Sin[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

```
Int[tan[d_.+e_.*x_]^m_*(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  1/(4^p*c^p)*Int[Tan[d+e*x]^m*(b+2*c*Cos[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

2: $\int \mathrm{Cot}[d+e\,x]^m \left(a+b\,\mathrm{Sin}[d+e\,x]^n+c\,\mathrm{Sin}[d+e\,x]^{2\,n}\right)^p \,dx$ when $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c=0\,\land\,p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2-4\,a\,c=0$, then $\partial_x \frac{\left(a+b\,F[x]+c\,F[x]^2\right)^p}{(b+2\,c\,F[x])^{2\,p}}=0$

Rule: If $\frac{m-1}{2}\notin\mathbb{Z}\,\land\,b^2-4\,a\,c=0\,\land\,p\notin\mathbb{Z}$, then

$$\int \mathrm{Cot}[d+e\,x]^m \left(a+b\,\mathrm{Sin}[d+e\,x]^n+c\,\mathrm{Sin}[d+e\,x]^{2\,n}\right)^p dx \to \frac{\left(a+b\,\mathrm{Sin}[d+e\,x]^n+c\,\mathrm{Sin}[d+e\,x]^{2\,n}\right)^p}{\left(b+2\,c\,\mathrm{Sin}[d+e\,x]^n\right)^{2\,p}}\int \mathrm{Cot}[d+e\,x]^m \left(b+2\,c\,\mathrm{Sin}[d+e\,x]^n\right)^{2\,p} dx$$

Program code:

```
Int[cot[d_.+e_.*x_]^m_*(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Sin[d+e*x]^n+c*Sin[d+e*x]^(2*n))^p/(b+2*c*Sin[d+e*x]^n)^(2*p)*Int[Cot[d+e*x]^m*(b+2*c*Sin[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

```
Int[tan[d_.+e_.*x_]^m_*(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_,x_Symbol] :=
  (a+b*Cos[d+e*x]^n+c*Cos[d+e*x]^(2*n))^p/(b+2*c*Cos[d+e*x]^n)^(2*p)*Int[Tan[d+e*x]^m*(b+2*c*Cos[d+e*x]^n)^(2*p),x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && EqQ[n2,2*n] && Not[IntegerQ[(m-1)/2]] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[p]]
```

2. $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m-1}{2}\notin\mathbb{Z}\,\wedge\,b^2-4\,a\,c\neq 0$

1: $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx$ when $\frac{m-1}{2}\notin\mathbb{Z}\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,\frac{n}{2}\in\mathbb{Z}\,\wedge\,p\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: $\text{Sin}[z]^2 == \frac{1}{1+\text{Cot}[z]^2}$

Basis: $\text{Cot}[d+e\,x]^m\,F\left[\text{Sin}[d+e\,x]^2\right] == -\frac{1}{e}\,\text{Subst}\left[\frac{x^m\,F\left[\frac{1}{1+x^2}\right]}{1+x^2},\,x,\,\text{Cot}[d+e\,x]\right]\partial_x\text{Cot}[d+e\,x]$

Rule: If $\frac{m-1}{2}\notin\mathbb{Z}\,\wedge\,b^2-4\,a\,c\neq 0\,\wedge\,\frac{n}{2}\in\mathbb{Z}\,\wedge\,p\in\mathbb{Z}$, then

$$\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx \to -\frac{1}{e}\,\text{Subst}\left[\int\frac{x^m\left(c+b\left(1+x^2\right)^{n/2}+a\left(1+x^2\right)^n\right)^p}{\left(1+x^2\right)^{n\,p+1}}dx,\,x,\,\text{Cot}[d+e\,x]\right]$$

Program code:

```
Int[cot[d_.+e_.*x_]^m_.*(a_+b_.*sin[d_.+e_.*x_]^n_+c_.*sin[d_.+e_.*x_]^n2_)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Cot[d+e*x],x]},
  -f^(m+1)/e*Subst[Int[x^m*ExpandToSum[c+b*(1+f^2*x^2)^(n/2)+a*(1+f^2*x^2)^n,x]^p/(1+f^2*x^2)^(n*p+1),x],x,Cot[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e,m},x] && EqQ[n2,2*n] && IntegerQ[n/2] && IntegerQ[p]
```

```
Int[tan[d_.+e_.*x_]^m_.*(a_+b_.*cos[d_.+e_.*x_]^n_+c_.*cos[d_.+e_.*x_]^n2_)^p_.,x_Symbol] :=
  Module[{f=FreeFactors[Tan[d+e*x],x]},
  f^(m+1)/e*Subst[Int[x^m*ExpandToSum[c+b*(1+f^2*x^2)^(n/2)+a*(1+f^2*x^2)^n,x]^p/(1+f^2*x^2)^(n*p+1),x],x,Tan[d+e*x]/f]] /;
FreeQ[{a,b,c,d,e,m},x] && EqQ[n2,2*n] && IntegerQ[n/2] && IntegerQ[p]
```

2: $\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p \, dx$ when $\frac{m}{2} \in \mathbb{Z} \land b^2-4\,a\,c \neq 0 \land (n \mid p) \in \mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\text{Cot}[z]^2 == \frac{1-\text{Sin}[z]^2}{\text{Sin}[z]^2}$

Rule: If $\frac{m}{2} \in \mathbb{Z} \land b^2-4\,a\,c \neq 0 \land (n \mid p) \in \mathbb{Z}$, then

$$\int \text{Cot}[d+e\,x]^m \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p dx \longrightarrow \int \text{ExpandTrig}\left[\frac{\left(1-\text{Sin}[d+e\,x]^2\right)^{m/2} \left(a+b\,\text{Sin}[d+e\,x]^n+c\,\text{Sin}[d+e\,x]^{2\,n}\right)^p}{\text{Sin}[d+e\,x]^m}, x\right] dx$$

Proeram code:

```
Int[cot[d_.+e_.*x_]^m_.*(a_.+b_.*sin[d_.+e_.*x_]^n_.+c_.*sin[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Int[ExpandTrig[(1-sin[d+e*x]^2)^(m/2)*(a+b*sin[d+e*x]^n+c*sin[d+e*x]^(2*n))^p/sin[d+e*x]^m,x],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[m/2] && NeQ[b^2-4*a*c,0] && IntegersQ[n,p]
```

```
Int[tan[d_.+e_.*x_]^m_.*(a_.+b_.*cos[d_.+e_.*x_]^n_.+c_.*cos[d_.+e_.*x_]^n2_.)^p_.,x_Symbol] :=
  Int[ExpandTrig[(1-cos[d+e*x]^2)^(m/2)*(a+b*cos[d+e*x]^n+c*cos[d+e*x]^(2*n))^p/cos[d+e*x]^m,x],x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[n2,2*n] && IntegerQ[m/2] && NeQ[b^2-4*a*c,0] && IntegersQ[n,p]
```

6. $\int (A + B \sin[d + e x]) (a + b \sin[d + e x] + c \sin[d + e x]^2)^n \, dx$

1. $\int (A + B \sin[d + e x]) (a + b \sin[d + e x] + c \sin[d + e x]^2)^n \, dx$ when $b^2 - 4 a c == 0$

1: $\int (A + B \sin[d + e x]) (a + b \sin[d + e x] + c \sin[d + e x]^2)^n \, dx$ when $b^2 - 4 a c == 0 \land n \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2 - 4 a c == 0$, then $a + b z + c z^2 == \frac{(b+2 c z)^2}{4 c}$

Rule: If $b^2 - 4 a c == 0 \land n \in \mathbb{Z}$, then

$$\int (A + B \sin[d + e x]) (a + b \sin[d + e x] + c \sin[d + e x]^2)^n \, dx \to \frac{1}{4^n c^n} \int (A + B \sin[d + e x]) (b + 2 c \sin[d + e x])^{2 n} \, dx$$

Program code:

```
Int[(A_+B_.*sin[d_.+e_.*x_])*(a_+b_.*sin[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  1/(4^n*c^n)*Int[(A+B*Sin[d+e*x])*(b+2*c*Sin[d+e*x])^(2*n),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2-4*a*c,0] && IntegerQ[n]
```

```
Int[(A_+B_.*cos[d_.+e_.*x_])*(a_+b_.*cos[d_.+e_.*x_]+c_.*cos[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  1/(4^n*c^n)*Int[(A+B*Cos[d+e*x])*(b+2*c*Cos[d+e*x])^(2*n),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2-4*a*c,0] && IntegerQ[n]
```

2: $\int (A + B \sin[d + e x]) \left(a + b \sin[d + e x] + c \sin[d + e x]^2\right)^n dx$ when $b^2 - 4 a c == 0 \land n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b^2 - 4 a c == 0$, then $\partial_x \frac{\left(a+b F[x]+c F[x]^2\right)^n}{(b+2 c F[x])^{2 n}} == 0$

Rule: If $b^2 - 4 a c == 0 \land n \notin \mathbb{Z}$, then

$$\int (A + B \sin[d + e x]) \left(a + b \sin[d + e x] + c \sin[d + e x]^2\right)^n dx \to \frac{\left(a + b \sin[d + e x] + c \sin[d + e x]^2\right)^n}{(b + 2 c \sin[d + e x])^{2 n}} \int (A + B \sin[d + e x]) (b + 2 c \sin[d + e x])^{2 n} dx$$

Program code:

```
Int[(A_+B_.*sin[d_.+e_.*x_])*(a_+b_.*sin[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  (a+b*Sin[d+e*x]+c*Sin[d+e*x]^2)^n/(b+2*c*Sin[d+e*x])^(2*n)*Int[(A+B*Sin[d+e*x])*(b+2*c*Sin[d+e*x])^(2*n),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[n]]
```

```
Int[(A_+B_.*cos[d_.+e_.*x_])*(a_+b_.*cos[d_.+e_.*x_]+c_.*cos[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  (a+b*Cos[d+e*x]+c*Cos[d+e*x]^2)^n/(b+2*c*Cos[d+e*x])^(2*n)*Int[(A+B*Cos[d+e*x])*(b+2*c*Cos[d+e*x])^(2*n),x] /;
FreeQ[{a,b,c,d,e,A,B},x] && EqQ[b^2-4*a*c,0] && Not[IntegerQ[n]]
```

2. $\int (A + B \operatorname{Sin}[d + e x]) \left(a + b \operatorname{Sin}[d + e x] + c \operatorname{Sin}[d + e x]^2\right)^n dx$ when $b^2 - 4 a c \neq 0$

1: $\int \frac{A + B \operatorname{Sin}[d + e x]}{a + b \operatorname{Sin}[d + e x] + c \operatorname{Sin}[d + e x]^2} dx$ when $b^2 - 4 a c \neq 0$

Derivation: Algebraic expansion

Basis: If $q = \sqrt{b^2 - 4 a c}$, then $\frac{A+B z}{a+b z+c z^2} == \left(B + \frac{b B-2 A c}{q}\right) \frac{1}{b+q+2 c z} + \left(B - \frac{b B-2 A c}{q}\right) \frac{1}{b-q+2 c z}$

Rule: If $b^2 - 4 a c \neq 0$, let $q = \sqrt{b^2 - 4 a c}$, then

$$\int \frac{A + B \operatorname{Sin}[d + e x]}{a + b \operatorname{Sin}[d + e x] + c \operatorname{Sin}[d + e x]^2} dx \to \left(B + \frac{b B - 2 A c}{q}\right) \int \frac{1}{b + q + 2 c \operatorname{Sin}[d + e x]} dx + \left(B - \frac{b B - 2 A c}{q}\right) \int \frac{1}{b - q + 2 c \operatorname{Sin}[d + e x]} dx$$

Program code:

```
Int[(A_+B_.*sin[d_.+e_.*x_])/(a_.+b_.*sin[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]^2),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  (B+(b*B-2*A*c)/q)*Int[1/(b+q+2*c*Sin[d+e*x]),x] +
  (B-(b*B-2*A*c)/q)*Int[1/(b-q+2*c*Sin[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0]
```

```
Int[(A_+B_.*cos[d_.+e_.*x_])/(a_.+b_.*cos[d_.+e_.*x_]+c_.*cos[d_.+e_.*x_]^2),x_Symbol] :=
  Module[{q=Rt[b^2-4*a*c,2]},
  (B+(b*B-2*A*c)/q)*Int[1/(b+q+2*c*Cos[d+e*x]),x] +
  (B-(b*B-2*A*c)/q)*Int[1/(b-q+2*c*Cos[d+e*x]),x]] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0]
```

2: $\int (A + B \operatorname{Sin}[d + e x]) \left(a + b \operatorname{Sin}[d + e x] + c \operatorname{Sin}[d + e x]^2\right)^n dx$ when $b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}$

$$\int (A + B \operatorname{Sin}[d + e x]) \left(a + b \operatorname{Sin}[d + e x] + c \operatorname{Sin}[d + e x]^2\right)^n dx \to \int \operatorname{ExpandTrig}\left[(A + B \operatorname{Sin}[d + e x]) \left(a + b \operatorname{Sin}[d + e x] + c \operatorname{Sin}[d + e x]^2\right)^n, x\right] dx$$

Program code:

```
Int[(A_+B_.*sin[d_.+e_.*x_])*(a_.+b_.*sin[d_.+e_.*x_]+c_.*sin[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  Int[ExpandTrig[(A+B*sin[d+e*x])*(a+b*sin[d+e*x]+c*sin[d+e*x]^2)^n,x],x] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && IntegerQ[n]
```

```
Int[(A_+B_.*cos[d_.+e_.*x_])*(a_.+b_.*cos[d_.+e_.*x_]+c_.*cos[d_.+e_.*x_]^2)^n_,x_Symbol] :=
  Int[ExpandTrig[(A+B*cos[d+e*x])*(a+b*cos[d+e*x]+c*cos[d+e*x]^2)^n,x],x] /;
FreeQ[{a,b,c,d,e,A,B},x] && NeQ[b^2-4*a*c,0] && IntegerQ[n]
```