Rules for integrands of the form $(c\, x)^m\, (a + b\, x^n)^p$

D: $\int (c\, x)^m \left(a_1 + b_1\, x^n\right)^p \left(a_2 + b_2\, x^n\right)^p dx$ when $a_2\, b_1 + a_1\, b_2 == 0 \land (p \in \mathbb{Z} \lor (a_1 > 0 \land a_2 > 0))$

Derivation: Algebraic simplification

Basis: If $a_2\, b_1 + a_1\, b_2 == 0 \land (p \in \mathbb{Z} \lor (a_1 > 0 \land a_2 > 0))$, then $\left(a_1 + b_1\, x^n\right)^p \left(a_2 + b_2\, x^n\right)^p == \left(a_1\, a_2 + b_1\, b_2\, x^{2\,n}\right)^p$

Rule: If $a_2\, b_1 + a_1\, b_2 == 0 \land (p \in \mathbb{Z} \lor (a_1 > 0 \land a_2 > 0))$, then

$$\int (c\, x)^m \left(a_1 + b_1\, x^n\right)^p \left(a_2 + b_2\, x^n\right)^p dx \longrightarrow \int (c\, x)^m \left(a_1\, a_2 + b_1\, b_2\, x^{2\,n}\right)^p dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  Int[(c*x)^m*(a1*a2+b1*b2*x^(2*n))^p,x] /;
FreeQ[{a1,b1,a2,b2,c,m,n,p},x] && EqQ[a2*b1+a1*b2,0] && (IntegerQ[p] || GtQ[a1,0] && GtQ[a2,0])
```

1. $\int x^m \left(a+b x^n\right)^p \, dx$ when $m == n-1$

1: $\int \frac{x^m}{a+b x^n} \, dx$ when $m == n-1$

Derivation: Integration by substitution and reciprocal rule for integration

Basis: If $m == n-1$, then $x^m F[x^n] == \frac{1}{n} F[x^n] \partial_x x^n$

Rule 1.1.3.2.1.1: If $m == n-1$, then

$$\int \frac{x^m}{a+b x^n} \, dx \longrightarrow \frac{1}{n} \text{Subst}\left[\int \frac{1}{a+b x} \, dx,\ x,\ x^n\right] \longrightarrow \frac{\text{Log}\left[a+b x^n\right]}{b n}$$

Program code:

```
Int[x_^m_./(a_+b_.*x_^n_),x_Symbol] :=
  Log[RemoveContent[a+b*x^n,x]]/(b*n) /;
FreeQ[{a,b,m,n},x] && EqQ[m,n-1]
```

2: $\int x^m \left(a+b x^n\right)^p \, dx$ when $m == n-1 \land p \neq -1$

Reference: G&R 2.110.4, CRC 88a with $m = n-1$

Derivation: Binomial recurrence 2a with $m = n-1$

Derivation: Integration by substitution and power rule for integration

Basis: If $m == n-1$, then $x^m F[x^n] == \frac{1}{n} F[x^n] \partial_x x^n$

Rule 1.1.3.2.1.2: If $m == n-1 \land p \neq -1$, then

$$\int x^m \left(a+b x^n\right)^p dx \longrightarrow \frac{1}{n} \text{Subst}\left[\int (a+b x)^p dx, x, x^n\right] \longrightarrow \frac{\left(a+b x^n\right)^{p+1}}{b n (p+1)}$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  (a+b*x^n)^(p+1)/(b*n*(p+1)) /;
FreeQ[{a,b,m,n,p},x] && EqQ[m,n-1] && NeQ[p,-1]
```

```
Int[x_^m_.*(a1_+b1_.*x_^n_.)^p_*(a2_+b2_.*x_^n_.)^p_,x_Symbol] :=
  (a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(2*b1*b2*n*(p+1)) /;
FreeQ[{a1,b1,a2,b2,m,n,p},x] && EqQ[a2*b1+a1*b2,0] && EqQ[m,2*n-1] && NeQ[p,-1]
```

2: $\int x^m \left(a+b x^n\right)^p dx$ when $p \in \mathbb{Z} \land n < 0$

Derivation: Algebraic expansion

Basis: If $p \in \mathbb{Z}$, then $\left(a+b x^n\right)^p == x^{n p} \left(b+a x^{-n}\right)^p$

Rule 1.1.3.2.2: If $p \in \mathbb{Z} \land n < 0$, then

$$\int x^m \left(a+b x^n\right)^p dx \longrightarrow \int x^{m+n p} \left(b+a x^{-n}\right)^p dx$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  Int[x^(m+n*p)*(b+a*x^(-n))^p,x] /;
FreeQ[{a,b,m,n},x] && IntegerQ[p] && NegQ[n]
```

3: $\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx$ when $\frac{m+1}{n}+p+1==0\ \land\ m\neq -1$

Reference: G&R 2.110.6, CRC 88c with $m+n\,p+n+1==0$

Derivation: Binomial recurrence 3b with $m+n\,p+n+1==0$

Rule 1.1.3.2.3: If $\frac{m+1}{n}+p+1==0\ \land\ m\neq -1$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx\ \longrightarrow\ \frac{(c\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}}{a\,c\,(m+1)}$$

Program code:

```
Int[(c_.*x_)^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a+b*x^n)^(p+1)/(a*c*(m+1)) /;
FreeQ[{a,b,c,m,n,p},x] && EqQ[(m+1)/n+p+1,0] && NeQ[m,-1]
```

```
Int[(c_.*x_)^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(a1*a2*c*(m+1)) /;
FreeQ[{a1,b1,a2,b2,c,m,n,p},x] && EqQ[a2*b1+a1*b2,0] && EqQ[(m+1)/(2*n)+p+1,0] && NeQ[m,-1]
```

4. $\int (c\,x)^m \left(a+b\,x^n\right)^p \,dx$ when $\frac{m+1}{n} \in \mathbb{Z}$

1: $\int x^m \left(a+b\,x^n\right)^p \,dx$ when $\frac{m+1}{n} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n} \in \mathbb{Z}$, then $x^m\,F[x^n] = \frac{1}{n}\,\text{Subst}\left[x^{\frac{m+1}{n}-1}\,F[x],\,x,\,x^n\right]\partial_x x^n$

Note: If $n \in \mathbb{Z}\ \land\ \frac{m+1}{n} \in \mathbb{Z}$, then $m \in \mathbb{Z}$, and $(c\,x)^m$ automatically evaluates to $c^m\,x^m$.

Rule 1.1.3.2.4.1: If $\frac{m+1}{n} \in \mathbb{Z}$, then

$$\int x^m \left(a+b\,x^n\right)^p \,dx \longrightarrow \frac{1}{n}\,\text{Subst}\left[\int x^{\frac{m+1}{n}-1}\left(a+b\,x\right)^p \,dx,\,x,\,x^n\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a+b*x)^p,x],x,x^n] /;
FreeQ[{a,b,m,n,p},x] && IntegerQ[Simplify[(m+1)/n]]
```

```
Int[x_^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  1/n*Subst[Int[x^(Simplify[(m+1)/n]-1)*(a1+b1*x)^p*(a2+b2*x)^p,x],x,x^n] /;
FreeQ[{a1,b1,a2,b2,m,n,p},x] && EqQ[a2*b1+a1*b2,0] && IntegerQ[Simplify[(m+1)/(2*n)]]
```

2: $\int (c\,x)^m\,(a+b\,x^n)^p\,dx$ when $\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m} == 0$

Rule 1.1.3.2.4.2: If $\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int (c\,x)^m\,(a+b\,x^n)^p\,dx \;\longrightarrow\; \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[(c_*x_)^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,m,n,p},x] && IntegerQ[Simplify[(m+1)/n]]
```

```
Int[(c_*x_)^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a1+b1*x^n)^p*(a2+b2*x^n)^p,x] /;
FreeQ[{a1,b1,a2,b2,c,m,n,p},x] && EqQ[a2*b1+a1*b2,0] && IntegerQ[Simplify[(m+1)/(2*n)]]
```

5: $\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx$ when $p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule 1.1.3.2.5: If $p\in\mathbb{Z}^+$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx \longrightarrow \int \text{ExpandIntegrand}\left[(c\,x)^m \left(a+b\,x^n\right)^p,\ x\right]dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_+b_.*x_^n_)^p_.,x_Symbol] :=
  Int[ExpandIntegrand[(c*x)^m*(a+b*x^n)^p,x],x] /;
FreeQ[{a,b,c,m,n},x] && IGtQ[p,0]
```

6. $\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx$ when $\frac{m+1}{n}+p+1\in\mathbb{Z}^-$

1: $\int x^m \left(a+b\,x^n\right)^p\,dx$ when $\frac{m+1}{n}+p+1\in\mathbb{Z}^-\land m\neq -1$

Reference: G&R 2.110.6, CRC 88c

Derivation: Binomial recurrence 3b

Note: This rule drives $\frac{m+1}{n}+p+1$ to 0 by incrementing m by n.

Rule 1.1.3.2.6.1: If $\frac{m+1}{n}+p+1\in\mathbb{Z}^-\land m\neq -1$, then

$$\int x^m \left(a+b\,x^n\right)^p\,dx \to \frac{x^{m+1}\left(a+b\,x^n\right)^{p+1}}{a\,(m+1)}-\frac{b\,(m+n\,(p+1)+1)}{a\,(m+1)}\int x^{m+n}\left(a+b\,x^n\right)^p\,dx$$

Program code:

```
Int[x_^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x^(m+1)*(a+b*x^n)^(p+1)/(a*(m+1)) -
  b*(m+n*(p+1)+1)/(a*(m+1))*Int[x^(m+n)*(a+b*x^n)^p,x] /;
FreeQ[{a,b,m,n,p},x] && ILtQ[Simplify[(m+1)/n+p+1],0] && NeQ[m,-1]
```

```
Int[x_^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  x^(m+1)*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(a1*a2*(m+1)) -
  b1*b2*(m+2*n*(p+1)+1)/(a1*a2*(m+1))*Int[x^(m+2*n)*(a1+b1*x^n)^p*(a2+b2*x^n)^p,x] /;
FreeQ[{a1,b1,a2,b2,m,n,p},x] && EqQ[a2*b1+a1*b2,0] && ILtQ[Simplify[(m+1)/(2*n)+p+1],0] && NeQ[m,-1]
```

2: $\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx$ when $\frac{m+1}{n}+p+1\in\mathbb{Z}^-\land p\neq-1$

Reference: G&R 2.110.2, CRC 88d

Derivation: Binomial recurrence 2b

Derivation: Integration by parts

Basis: $x^m\ \left(a+b\,x^n\right)^p == x^{m+n\,p+n+1}\ \frac{\left(a+b\,x^n\right)^p}{x^{n\,(p+1)+1}}$

Basis: $\int \frac{\left(a+b\,x^n\right)^p}{x^{n\,(p+1)+1}}\,dx == -\frac{\left(a+b\,x^n\right)^{p+1}}{x^{n\,(p+1)}\ a\,n\,(p+1)}$

Note: This rule drives $\frac{m+1}{n}+p+1$ to 0 by incrementing p by 1.

Rule 1.1.3.2.6.2: If $\frac{m+1}{n}+p+1\in\mathbb{Z}^-\land\ p\neq-1$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx \longrightarrow -\frac{(c\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}}{a\,c\,n\,(p+1)}+\frac{m+n\,(p+1)+1}{a\,n\,(p+1)}\int (c\,x)^m \left(a+b\,x^n\right)^{p+1}\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  -(c*x)^(m+1)*(a+b*x^n)^(p+1)/(a*c*n*(p+1)) +
  (m+n*(p+1)+1)/(a*n*(p+1))*Int[(c*x)^m*(a+b*x^n)^(p+1),x] /;
FreeQ[{a,b,c,m,n,p},x] && ILtQ[Simplify[(m+1)/n+p+1],0] && NeQ[p,-1]
```

```
Int[(c_.*x_)^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  -(c*x)^(m+1)*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(2*a1*a2*c*n*(p+1)) +
  (m+2*n*(p+1)+1)/(2*a1*a2*n*(p+1))*Int[(c*x)^m*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1),x] /;
FreeQ[{a1,b1,a2,b2,c,m,n,p},x] && EqQ[a2*b1+a1*b2,0] && ILtQ[Simplify[(m+1)/(2*n)+p+1],0] && NeQ[p,-1]
```

7. $\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx$ when $n \in \mathbb{Z}$

1. $\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx$ when $n \in \mathbb{Z}^+$

1: $\int x^m \left(a+b\,x^n\right)^p\,dx$ when $n \in \mathbb{Z}^+ \land m \in \mathbb{Z} \land \text{GCD}[m+1,\ n] \neq 1$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z} \land m \in \mathbb{Z}$, let $k = \text{GCD}[m+1,\ n]$, then $x^m\,F[x^n] = \frac{1}{k}\,\text{Subst}\left[x^{\frac{m+1}{k}-1}\,F\left[x^{n/k}\right],\ x,\ x^k\right]\,\partial_x x^k$

Rule 1.1.3.2.7.1.1: If $n \in \mathbb{Z}^+ \land m \in \mathbb{Z}$, let $k = \text{GCD}[m+1,\ n]$, if $k \neq 1$, then

$$\int x^m \left(a+b\,x^n\right)^p\,dx \longrightarrow \frac{1}{k}\,\text{Subst}\left[\int x^{\frac{m+1}{k}-1} \left(a+b\,x^{n/k}\right)^p\,dx,\ x,\ x^k\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{k=GCD[m+1,n]},
  1/k*Subst[Int[x^((m+1)/k-1)*(a+b*x^(n/k))^p,x],x,x^k] /;
 k≠1] /;
FreeQ[{a,b,p},x] && IGtQ[n,0] && IntegerQ[m]
```

```
Int[x_^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  With[{k=GCD[m+1,2*n]},
  1/k*Subst[Int[x^((m+1)/k-1)*(a1+b1*x^(n/k))^p*(a2+b2*x^(n/k))^p,x],x,x^k] /;
 k≠1] /;
FreeQ[{a1,b1,a2,b2,p},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && IntegerQ[m]
```

2. $\int (c\,x)^m \left(a+b\,x^n\right)^p dx$ when $n \in \mathbb{Z}^+ \land p > 0$

1: $\int (c\,x)^m \left(a+b\,x^n\right)^p dx$ when $n \in \mathbb{Z}^+ \land p > 0 \land m < -1$

Reference: G&R 2.110.3

Derivation: Binomial recurrence 1a

Derivation: Integration by parts

Rule 1.1.3.2.7.1.2.1: If $n \in \mathbb{Z}^+ \land p > 0 \land m < -1$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p dx \to \frac{(c\,x)^{m+1} \left(a+b\,x^n\right)^p}{c\,(m+1)} - \frac{b\,n\,p}{c^n\,(m+1)} \int (c\,x)^{m+n} \left(a+b\,x^n\right)^{p-1} dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a+b*x^n)^p/(c*(m+1)) -
  b*n*p/(c^n*(m+1))*Int[(c*x)^(m+n)*(a+b*x^n)^(p-1),x] /;
FreeQ[{a,b,c},x] && IGtQ[n,0] && GtQ[p,0] && LtQ[m,-1] && Not[ILtQ[(m+n*p+n+1)/n,0]] &&
  IntBinomialQ[a,b,c,n,m,p,x]
```

```
Int[(c_.*x_)^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a1+b1*x^n)^p*(a2+b2*x^n)^p/(c*(m+1)) -
  2*b1*b2*n*p/(c^(2*n)*(m+1))*Int[(c*x)^(m+2*n)*(a1+b1*x^n)^(p-1)*(a2+b2*x^n)^(p-1),x] /;
FreeQ[{a1,b1,a2,b2,c,m},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && GtQ[p,0] && LtQ[m,-1] && NeQ[m+2*n*p+1,0] &&
  IntBinomialQ[a1*a2,b1*b2,c,2*n,m,p,x]
```

2: $\int (c\,x)^m \left(a+b\,x^n\right)^p dx$ when $n \in \mathbb{Z}^+ \land p > 0 \land m+n\,p+1 \neq 0$

Reference: G&R 2.110.1, CRC 88b

Derivation: Binomial recurrence 1b

Derivation: Inverted integration by parts

Rule 1.1.3.2.7.1.2.2: If $n \in \mathbb{Z}^+ \land p > 0 \land m+n\,p+1 \neq 0$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p dx \longrightarrow \frac{(c\,x)^{m+1} \left(a+b\,x^n\right)^p}{c\,(m+n\,p+1)} + \frac{a\,n\,p}{m+n\,p+1} \int (c\,x)^m \left(a+b\,x^n\right)^{p-1} dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a+b*x^n)^p/(c*(m+n*p+1)) +
  a*n*p/(m+n*p+1)*Int[(c*x)^m*(a+b*x^n)^(p-1),x] /;
FreeQ[{a,b,c,m},x] && IGtQ[n,0] && GtQ[p,0] && NeQ[m+n*p+1,0] && IntBinomialQ[a,b,c,n,m,p,x]
```

```
Int[(c_.*x_)^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a1+b1*x^n)^p*(a2+b2*x^n)^p/(c*(m+2*n*p+1)) +
  2*a1*a2*n*p/(m+2*n*p+1)*Int[(c*x)^m*(a1+b1*x^n)^(p-1)*(a2+b2*x^n)^(p-1),x] /;
FreeQ[{a1,b1,a2,b2,c,m},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && GtQ[p,0] && NeQ[m+2*n*p+1,0] && IntBinomialQ[a1*a2,b1*b2,c,2*n,m,
```

3. $\int (c\,x)^m\,(a+b\,x^n)^p\,dx$ when $n\in\mathbb{Z}^+\land p<-1$

1. $\int \frac{x^m}{(a+b\,x^4)^{5/4}}\,dx$ when $\frac{b}{a}>0\land\frac{m-2}{4}\in\mathbb{Z}$

1: $\int \frac{x^2}{(a+b\,x^4)^{5/4}}\,dx$ when $\frac{b}{a}>0$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{x\left(1+\frac{a}{b\,x^4}\right)^{1/4}}{(a+b\,x^4)^{1/4}}==0$

Rule 1.1.3.2.7.1.3.1.1: If $\frac{b}{a}>0$, then

$$\int \frac{x^2}{(a+b\,x^4)^{5/4}}\,dx \longrightarrow \frac{x\left(1+\frac{a}{b\,x^4}\right)^{1/4}}{b\,(a+b\,x^4)^{1/4}}\int\frac{1}{x^3\left(1+\frac{a}{b\,x^4}\right)^{5/4}}\,dx$$

Program code:

```
Int[x_^2/(a_+b_.*x_^4)^(5/4),x_Symbol] :=
  x*(1+a/(b*x^4))^(1/4)/(b*(a+b*x^4)^(1/4))*Int[1/(x^3*(1+a/(b*x^4))^(5/4)),x] /;
FreeQ[{a,b},x] && PosQ[b/a]
```

2: $\int \frac{x^m}{\left(a+b\,x^4\right)^{5/4}}\,dx$ when $\frac{b}{a}>0\,\land\,\frac{m-2}{4}\in\mathbb{Z}^+$

Reference: G&R 2.110.5, CRC 88a

Derivation: Binomial recurrence 3a

Derivation: Inverted integration by parts

Rule 1.1.3.2.7.1.3.1.2: If $\frac{b}{a}>0\,\land\,\frac{m-2}{4}\in\mathbb{Z}^+$, then

$$\int \frac{x^m}{\left(a+b\,x^4\right)^{5/4}}\,dx\,\longrightarrow\,\frac{x^{m-3}}{b\,(m-4)\,\left(a+b\,x^4\right)^{1/4}}-\frac{a\,(m-3)}{b\,(m-4)}\int \frac{x^{m-4}}{\left(a+b\,x^4\right)^{5/4}}\,dx$$

Program code:

```
Int[x_^m_/(a_+b_.*x_^4)^(5/4),x_Symbol] :=
  x^(m-3)/(b*(m-4)*(a+b*x^4)^(1/4)) - a*(m-3)/(b*(m-4))*Int[x^(m-4)/(a+b*x^4)^(5/4),x] /;
FreeQ[{a,b},x] && PosQ[b/a] && IGtQ[(m-2)/4,0]
```

3: $\int \frac{x^m}{\left(a+b\,x^4\right)^{5/4}}\,dx$ when $\frac{b}{a}>0\,\land\,\frac{m-2}{4}\in\mathbb{Z}^-$

Reference: G&R 2.110.6, CRC 88c

Derivation: Binomial recurrence 3b

Derivation: Integration by parts

Rule 1.1.3.2.7.1.3.1.3: If $\frac{b}{a}>0\,\land\,\frac{m-2}{4}\in\mathbb{Z}^-$, then

$$\int \frac{x^m}{\left(a+b\,x^4\right)^{5/4}}\,dx \longrightarrow \frac{x^{m+1}}{a\,(m+1)\,\left(a+b\,x^4\right)^{1/4}}-\frac{b\,m}{a\,(m+1)}\int \frac{x^{m+4}}{\left(a+b\,x^4\right)^{5/4}}\,dx$$

Program code:

```
Int[x_^m_/(a_+b_.*x_^4)^(5/4),x_Symbol] :=
  x^(m+1)/(a*(m+1)*(a+b*x^4)^(1/4)) - b*m/(a*(m+1))*Int[x^(m+4)/(a+b*x^4)^(5/4),x] /;
FreeQ[{a,b},x] && PosQ[b/a] && ILtQ[(m-2)/4,0]
```

2. $\int \frac{(c\,x)^m}{\left(a+b\,x^2\right)^{5/4}}\,dx$ when $\frac{b}{a}>0\,\land\,2\,m\in\mathbb{Z}$

1: $\int \frac{\sqrt{c\,x}}{\left(a+b\,x^2\right)^{5/4}}\,dx$ when $\frac{b}{a}>0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{c\,x}\,\left(1+\frac{a}{b\,x^2}\right)^{1/4}}{\left(a+b\,x^2\right)^{1/4}} == 0$

Rule 1.1.3.2.7.1.3.2.1: If $\frac{b}{a}>0$, then

$$\int \frac{\sqrt{c\,x}}{\left(a+b\,x^2\right)^{5/4}}\,dx \to \frac{\sqrt{c\,x}\,\left(1+\frac{a}{b\,x^2}\right)^{1/4}}{b\,\left(a+b\,x^2\right)^{1/4}}\int \frac{1}{x^2\,\left(1+\frac{a}{b\,x^2}\right)^{5/4}}\,dx$$

Program code:

```
Int[Sqrt[c_.*x_]/(a_+b_.*x_^2)^(5/4),x_Symbol] :=
  Sqrt[c*x]*(1+a/(b*x^2))^(1/4)/(b*(a+b*x^2)^(1/4))*Int[1/(x^2*(1+a/(b*x^2))^(5/4)),x] /;
FreeQ[{a,b,c},x] && PosQ[b/a]
```

2: $\int \frac{(c\,x)^m}{\left(a+b\,x^2\right)^{5/4}}\,dx$ when $\frac{b}{a}>0\,\land\,2\,m\in\mathbb{Z}\,\land\,m>\frac{3}{2}$

Reference: G&R 2.110.5, CRC 88a

Derivation: Binomial recurrence 3a

Derivation: Inverted integration by parts

Rule 1.1.3.2.7.1.3.2.2: If $\frac{b}{a}>0\,\land\,2\,m\in\mathbb{Z}\,\land\,m>\frac{3}{2}$, then

$$\int \frac{(c\,x)^m}{\left(a+b\,x^2\right)^{5/4}}\,dx \to \frac{2\,c\,(c\,x)^{m-1}}{b\,(2\,m-3)\,\left(a+b\,x^2\right)^{1/4}}-\frac{2\,a\,c^2\,(m-1)}{b\,(2\,m-3)}\int \frac{(c\,x)^{m-2}}{\left(a+b\,x^2\right)^{5/4}}\,dx$$

Program code:

```
Int[(c_.*x_)^m_/(a_+b_.*x_^2)^(5/4),x_Symbol] :=
  2*c*(c*x)^(m-1)/(b*(2*m-3)*(a+b*x^2)^(1/4)) - 2*a*c^2*(m-1)/(b*(2*m-3))*Int[(c*x)^(m-2)/(a+b*x^2)^(5/4),x] /;
FreeQ[{a,b,c},x] && PosQ[b/a] && IntegerQ[2*m] && GtQ[m,3/2]
```

3: $\int \frac{(c\,x)^m}{\left(a+b\,x^2\right)^{5/4}}\,dx$ when $\frac{b}{a}>0\,\land\,2\,m\in\mathbb{Z}\,\land\,m<-1$

Reference: G&R 2.110.6, CRC 88c

Derivation: Binomial recurrence 3b

Derivation: Integration by parts

Rule 1.1.3.2.7.1.3.2.3: If $\frac{b}{a}>0\,\land\,2\,m\in\mathbb{Z}\,\land\,m<-1$, then

$$\int \frac{(c\,x)^m}{\left(a+b\,x^2\right)^{5/4}}\,dx \longrightarrow \frac{(c\,x)^{m+1}}{a\,c\,(m+1)\,\left(a+b\,x^2\right)^{1/4}}-\frac{b\,(2\,m+1)}{2\,a\,c^2\,(m+1)}\int \frac{(c\,x)^{m+2}}{\left(a+b\,x^2\right)^{5/4}}\,dx$$

Program code:

```
Int[(c_.*x_)^m_/(a_+b_.*x_^2)^(5/4),x_Symbol] :=
  (c*x)^(m+1)/(a*c*(m+1)*(a+b*x^2)^(1/4)) - b*(2*m+1)/(2*a*c^2*(m+1))*Int[(c*x)^(m+2)/(a+b*x^2)^(5/4),x] /;
FreeQ[{a,b,c},x] && PosQ[b/a] && IntegerQ[2*m] && LtQ[m,-1]
```

3: $\int \frac{x^2}{\left(a+b\,x^4\right)^{5/4}}\,dx$ when $\frac{b}{a} \not> 0$

Reference: G&R 2.110.4

Derivation: Binomial recurrence 2a

Derivation: Integration by parts

Rule 1.1.3.2.7.1.3.3: If $\frac{b}{a} \not> 0$, then

$$\int \frac{x^2}{\left(a+b\,x^4\right)^{5/4}}\,dx \longrightarrow -\frac{1}{b\,x\,\left(a+b\,x^4\right)^{1/4}} - \frac{1}{b}\int \frac{1}{x^2\,\left(a+b\,x^4\right)^{1/4}}\,dx$$

Program code:

```
Int[x_^2/(a_+b_.*x_^4)^(5/4),x_Symbol] :=
  -1/(b*x*(a+b*x^4)^(1/4)) - 1/b*Int[1/(x^2*(a+b*x^4)^(1/4)),x] /;
FreeQ[{a,b},x] && NegQ[b/a]
```

4: $\int (c\,x)^m \left(a+b\,x^n\right)^p dx$ when $n \in \mathbb{Z}^+ \land p < -1 \land m+1 > n$

Reference: G&R 2.110.4

Derivation: Binomial recurrence 2a

Derivation: Integration by parts

Basis: $x^m \left(a+b\,x^n\right)^p == x^{m-n+1} \left(a+b\,x^n\right)^p x^{n-1}$

Basis: $\int \left(a+b\,x^n\right)^p x^{n-1}\,dx == \frac{\left(a+b\,x^n\right)^{p+1}}{b\,n\,(p+1)}$

Rule 1.1.3.2.7.1.3.4: If $n \in \mathbb{Z}^+ \land p < -1 \land m+1 > n$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p dx \rightarrow \frac{c^{n-1}\,(c\,x)^{m-n+1}\left(a+b\,x^n\right)^{p+1}}{b\,n\,(p+1)} - \frac{c^n\,(m-n+1)}{b\,n\,(p+1)} \int (c\,x)^{m-n} \left(a+b\,x^n\right)^{p+1} dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  c^(n-1)*(c*x)^(m-n+1)*(a+b*x^n)^(p+1)/(b*n*(p+1)) -
  c^n*(m-n+1)/(b*n*(p+1))*Int[(c*x)^(m-n)*(a+b*x^n)^(p+1),x] /;
FreeQ[{a,b,c},x] && IGtQ[n,0] && LtQ[p,-1] && GtQ[m+1,n] && Not[ILtQ[(m+n*(p+1)+1)/n,0]] && IntBinomialQ[a,b,c,n,m,p,x]
```

```
(* Int[(c_.*x_)^m_.*u_^p_*v_^p_,x_Symbol] :=
  With[{a=BinomialParts[u,x][[1]],b=BinomialParts[u,x][[2]],n=BinomialParts[u,x][[3]]},
  c^(n-1)*(c*x)^(m-n+1)*u^(p+1)*v^(p+1)/(b*n*(p+1)) -
  c^n*(m-n+1)/(b*n*(p+1))*Int[(c*x)^(m-n)*u^(p+1)*v^(p+1),x] /;
 IGtQ[n,0] && m+1>n && Not[ILtQ[(m+n*(p+1)+1)/n,0]] &&
   IntBinomialQ[a,b,c,n,m,p,x]] /;
FreeQ[c,x] && BinomialQ[u*v,x] && LtQ[p,-1] *)
```

```
Int[(c_.*x_)^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  c^(2*n-1)*(c*x)^(m-2*n+1)*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(2*b1*b2*n*(p+1)) -
  c^(2*n)*(m-2*n+1)/(2*b1*b2*n*(p+1))*Int[(c*x)^(m-2*n)*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1),x] /;
FreeQ[{a1,b1,a2,b2,c},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && LtQ[p,-1] && m+1>2*n &&
  Not[ILtQ[(m+2*n*(p+1)+1)/(2*n),0]] && IntBinomialQ[a1*a2,b1*b2,c,2*n,m,p,x]
```

5: $\int (c\,x)^m \left(a+b\,x^n\right)^p dx$ when $n \in \mathbb{Z}^+ \wedge p < -1$

Reference: G&R 2.110.2, CRC 88d

Derivation: Binomial recurrence 2b

Derivation: Integration by parts

Basis: $x^m \left(a+b\,x^n\right)^p == x^{m+n\,p+n+1} \frac{\left(a+b\,x^n\right)^p}{x^{n\,(p+1)+1}}$

Basis: $\int \frac{\left(a+b\,x^n\right)^p}{x^{n\,(p+1)+1}}\,dx == -\frac{\left(a+b\,x^n\right)^{p+1}}{x^{n\,(p+1)}\,a\,n\,(p+1)}$

Rule 1.1.3.2.7.1.3.5: If $n \in \mathbb{Z}^+ \wedge p < -1$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p dx \longrightarrow -\frac{(c\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}}{a\,c\,n\,(p+1)} + \frac{m+n\,(p+1)+1}{a\,n\,(p+1)} \int (c\,x)^m \left(a+b\,x^n\right)^{p+1} dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  -(c*x)^(m+1)*(a+b*x^n)^(p+1)/(a*c*n*(p+1)) +
  (m+n*(p+1)+1)/(a*n*(p+1))*Int[(c*x)^m*(a+b*x^n)^(p+1),x] /;
FreeQ[{a,b,c,m},x] && IGtQ[n,0] && LtQ[p,-1] && IntBinomialQ[a,b,c,n,m,p,x]
```

```
Int[(c_.*x_)^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  -(c*x)^(m+1)*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(2*a1*a2*c*n*(p+1)) +
  (m+2*n*(p+1)+1)/(2*a1*a2*n*(p+1))*Int[(c*x)^m*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1),x] /;
FreeQ[{a1,b1,a2,b2,c,m},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && LtQ[p,-1] && IntBinomialQ[a1*a2,b1*b2,c,2*n,m,p,x]
```

4. $\int \frac{x^m}{a + b x^n} \, dx$ when $n \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+$

    1. $\int \frac{x^m}{a + b x^n} \, dx$ when $n \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+ \land m < n - 1$

        1. $\int \frac{x^m}{a + b x^n} \, dx$ when $\frac{n-1}{2} \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+ \land m < n - 1$

            1: $\int \frac{x}{a + b x^3} \, dx$

Reference: G&R 2.126.2, CRC 75

Derivation: Algebraic expansion

Basis: $\frac{x}{a+b x^3} == -\frac{1}{3 a^{1/3} b^{1/3} \left(a^{1/3}+b^{1/3} x\right)} + \frac{a^{1/3}+b^{1/3} x}{3 a^{1/3} b^{1/3} \left(a^{2/3}-a^{1/3} b^{1/3} x+b^{2/3} x^2\right)}$

Rule 1.1.3.2.7.1.4.1.1.1:

$$\int \frac{x}{a + b x^3} \, dx \longrightarrow -\frac{1}{3 a^{1/3} b^{1/3}} \int \frac{1}{a^{1/3} + b^{1/3} x} \, dx + \frac{1}{3 a^{1/3} b^{1/3}} \int \frac{a^{1/3} + b^{1/3} x}{a^{2/3} - a^{1/3} b^{1/3} x + b^{2/3} x^2} \, dx$$

Program code:

```
Int[x_/(a_+b_.*x_^3),x_Symbol] :=
  -1/(3*Rt[a,3]*Rt[b,3])*Int[1/(Rt[a,3]+Rt[b,3]*x),x] +
  1/(3*Rt[a,3]*Rt[b,3])*Int[(Rt[a,3]+Rt[b,3]*x)/(Rt[a,3]^2-Rt[a,3]*Rt[b,3]*x+Rt[b,3]^2*x^2),x] /;
FreeQ[{a,b},x]
```

2. $\int \frac{x^m}{a+b x^5} \, dx$ when $m \in \mathbb{Z}^+ \land m<4$

1: $\int \frac{x^m}{a+b x^5} \, dx$ when $m \in \mathbb{Z}^+ \land m<4 \land \frac{a}{b}>0$

Derivation: Algebraic expansion

Basis: If $m \in \mathbb{Z} \land 0 \le m < 5$, let $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/5}$, then

$$\frac{x^m}{a+b x^5} == \frac{(-1)^m r^{m+1}}{5 a s^m (r+s x)} + \frac{2 r^{m+1}}{5 a s^m} \frac{r \operatorname{Cos}\left[\frac{m \pi}{5}\right] - s \operatorname{Cos}\left[\frac{(m+1) \pi}{5}\right] x}{r^2 - \frac{1}{2}\left(1+\sqrt{5}\right) r s x + s^2 x^2} + \frac{2 r^{m+1}}{5 a s^m} \frac{r \operatorname{Cos}\left[\frac{3 m \pi}{5}\right] - s \operatorname{Cos}\left[\frac{3 (m+1) \pi}{5}\right] x}{r^2 - \frac{1}{2}\left(1-\sqrt{5}\right) r s x + s^2 x^2}$$

Note: This rule not necessary for host systems that automatically simplify $\operatorname{Cos}\left[\frac{k \pi}{5}\right]$ to radicals when k is an integer.

Rule 1.1.3.2.7.1.4.1.1.2.1: If $m \in \mathbb{Z}^+ \land m < 4 \land \frac{a}{b} > 0$, let $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/5}$, then

$$\int \frac{x^m}{a+b x^5} \, dx \to \frac{(-1)^m r^{m+1}}{5 a s^m} \int \frac{1}{r+s x} \, dx + \frac{2 r^{m+1}}{5 a s^m} \int \frac{r \operatorname{Cos}\left[\frac{m \pi}{5}\right] - s \operatorname{Cos}\left[\frac{(m+1) \pi}{5}\right] x}{r^2 - \frac{1}{2}\left(1+\sqrt{5}\right) r s x + s^2 x^2} \, dx + \frac{2 r^{m+1}}{5 a s^m} \int \frac{r \operatorname{Cos}\left[\frac{3 m \pi}{5}\right] - s \operatorname{Cos}\left[\frac{3 (m+1) \pi}{5}\right] x}{r^2 - \frac{1}{2}\left(1-\sqrt{5}\right) r s x + s^2 x^2} \, dx$$

Program code:

```
(* Int[x_^m_./(a_+b_.*x_^5),x_Symbol] :=
  With[{r=Numerator[Rt[a/b,5]], s=Denominator[Rt[a/b,5]]},
  (-1)^m*r^(m+1)/(5*a*s^m)*Int[1/(r+s*x),x] +
   2*r^(m+1)/(5*a*s^m)*Int[(r*Cos[m*Pi/5]-s*Cos[(m+1)*Pi/5]*x)/(r^2-1/2*(1+Sqrt[5])*r*s*x+s^2*x^2),x] +
  2*r^(m+1)/(5*a*s^m)*Int[(r*Cos[3*m*Pi/5]-s*Cos[3*(m+1)*Pi/5]*x)/(r^2-1/2*(1-Sqrt[5])*r*s*x+s^2*x^2),x]] /;
FreeQ[{a,b},x] && IGtQ[m,0] && LtQ[m,4] && PosQ[a/b] *)
```

2: $\int \frac{x^m}{a+b x^5} \, dx$ when $m \in \mathbb{Z}^+ \land m<4 \land \frac{a}{b} \not> 0$

Derivation: Algebraic expansion

Basis: If $m \in \mathbb{Z} \land 0 \le m < 5$, let $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/5}$, then

$$\frac{x^m}{a+b\,x^5} == \frac{r^{m+1}}{5\,a\,s^m\,(r-s\,x)} + \frac{2\,(-1)^m\,r^{m+1}}{5\,a\,s^m}\,\frac{r\,\mathrm{Cos}\left[\frac{m\,\pi}{5}\right]+s\,\mathrm{Cos}\left[\frac{(m+1)\,\pi}{5}\right]x}{r^2+\frac{1}{2}\left(1+\sqrt{5}\right)r\,s\,x+s^2\,x^2} + \frac{2\,(-1)^m\,r^{m+1}}{5\,a\,s^m}\,\frac{r\,\mathrm{Cos}\left[\frac{3\,m\,\pi}{5}\right]+s\,\mathrm{Cos}\left[\frac{3\,(m+1)\,\pi}{5}\right]x}{r^2+\frac{1}{2}\left(1-\sqrt{5}\right)r\,s\,x+s^2\,x^2}$$

Note: This rule not necessary for host systems that automatically simplify $\mathrm{Cos}\left[\frac{k\,\pi}{5}\right]$ to radicals when k is an integer.

Rule 1.1.3.2.7.1.4.1.1.2.1: If $m \in \mathbb{Z}^+ \land m < 4 \land \frac{a}{b} \not> 0$, let $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/5}$, then

$$\int \frac{x^m}{a+b\,x^5}\,dx \to \frac{r^{m+1}}{5\,a\,s^m}\int \frac{1}{r-s\,x}\,dx + \frac{2\,(-1)^m\,r^{m+1}}{5\,a\,s^m}\int \frac{r\,\mathrm{Cos}\left[\frac{m\,\pi}{5}\right]+s\,\mathrm{Cos}\left[\frac{(m+1)\,\pi}{5}\right]x}{r^2+\frac{1}{2}\left(1+\sqrt{5}\right)r\,s\,x+s^2\,x^2}\,dx + \frac{2\,(-1)^m\,r^{m+1}}{5\,a\,s^m}\int \frac{r\,\mathrm{Cos}\left[\frac{3\,m\,\pi}{5}\right]+s\,\mathrm{Cos}\left[\frac{3\,(m+1)\,\pi}{5}\right]x}{r^2+\frac{1}{2}\left(1-\sqrt{5}\right)r\,s\,x+s^2\,x^2}\,dx$$

Program code:

```
(* Int[x_^m_./(a_+b_.*x_^5),x_Symbol] :=
  With[{r=Numerator[Rt[-a/b,5]], s=Denominator[Rt[-a/b,5]]},
  (r^(m+1)/(5*a*s^m))*Int[1/(r-s*x),x] +
  2*(-1)^m*r^(m+1)/(5*a*s^m)*Int[(r*Cos[m*Pi/5]+s*Cos[(m+1)*Pi/5]*x)/(r^2+1/2*(1+Sqrt[5])*r*s*x+s^2*x^2),x] +
  2*(-1)^m*r^(m+1)/(5*a*s^m)*Int[(r*Cos[3*m*Pi/5]+s*Cos[3*(m+1)*Pi/5]*x)/(r^2+1/2*(1-Sqrt[5])*r*s*x+s^2*x^2),x]] /;
FreeQ[{a,b},x] && IGtQ[m,0] && LtQ[m,4] && NegQ[a/b] *)
```

3. $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n-1}{2} \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+ \land m < n-1 \land n > 3$

1: $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n-1}{2} \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+ \land m < n-1 \land \frac{a}{b} > 0$

Derivation: Algebraic expansion

Basis: If $\frac{n-1}{2} \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+ \land m < n-1$, let $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/n}$, then

$$\frac{z^m}{a+b\,z^n} == -\frac{(-r)^{m+1}}{a\,n\,s^m\,(r+s\,z)} + \frac{2\,r^{m+1}}{a\,n\,s^m}\sum_{k=1}^{\frac{n-1}{2}} \frac{r\,\mathrm{Cos}\left[\frac{(2\,k-1)\,m\,\pi}{n}\right]-s\,\mathrm{Cos}\left[\frac{(2\,k-1)\,(m+1)\,\pi}{n}\right]z}{r^2-2\,r\,s\,\mathrm{Cos}\left[\frac{(2\,k-1)\,\pi}{n}\right]z+s^2\,z^2}$$

Rule 1.1.3.2.7.1.4.1.1.3.1: If $\frac{n-1}{2} \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+ \land m < n-1 \land \frac{a}{b} > 0$, let $\frac{r}{s} = \left(\frac{a}{b}\right)^{1/n}$, then

$$\int \frac{x^m}{a+b\,x^n}\,dx \to -\frac{(-r)^{m+1}}{a\,n\,s^m}\int \frac{1}{r+s\,x}\,dx+\frac{2\,r^{m+1}}{a\,n\,s^m}\sum_{k=1}^{\frac{n-1}{2}}\int \frac{r\,\mathrm{Cos}\left[\frac{(2\,k-1)\,m\,\pi}{n}\right]-s\,\mathrm{Cos}\left[\frac{(2\,k-1)\,(m+1)\,\pi}{n}\right]\,x}{r^2-2\,r\,s\,\mathrm{Cos}\left[\frac{(2\,k-1)\,\pi}{n}\right]\,x+s^2\,x^2}\,dx$$

Program code:

```
Int[x_^m_./(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,n]], s=Denominator[Rt[a/b,n]], k, u},
  u=Int[(r*Cos[(2*k-1)*m*Pi/n]-s*Cos[(2*k-1)*(m+1)*Pi/n]*x)/(r^2-2*r*s*Cos[(2*k-1)*Pi/n]*x+s^2*x^2),x];
  -(-r)^(m+1)/(a*n*s^m)*Int[1/(r+s*x),x] + Dist[2*r^(m+1)/(a*n*s^m),Sum[u,{k,1,(n-1)/2}],x]] /;
FreeQ[{a,b},x] && IGtQ[(n-1)/2,0] && IGtQ[m,0] && LtQ[m,n-1] && PosQ[a/b]
```

2: $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n-1}{2}\in\mathbb{Z}^+\wedge m\in\mathbb{Z}^+\wedge m<n-1\wedge \frac{a}{b}\ngtr 0$

Derivation: Algebraic expansion

Basis: If $\frac{n-1}{2}\in\mathbb{Z}^+\wedge m\in\mathbb{Z}^+\wedge m<n-1$, let $\frac{r}{s}=\left(-\frac{a}{b}\right)^{1/n}$, then

$$\frac{z^m}{a+b\,z^n}==\frac{r^{m+1}}{a\,n\,s^m\,(r-s\,z)}-\frac{2\,(-r)^{m+1}}{a\,n\,s^m}\sum_{k=1}^{\frac{n-1}{2}}\frac{r\,\mathrm{Cos}\left[\frac{(2\,k-1)\,m\,\pi}{n}\right]+s\,\mathrm{Cos}\left[\frac{(2\,k-1)\,(m+1)\,\pi}{n}\right]\,z}{r^2+2\,r\,s\,\mathrm{Cos}\left[\frac{(2\,k-1)\,\pi}{n}\right]\,z+s^2\,z^2}$$

Rule 1.1.3.2.7.1.4.1.1.3.2: If $\frac{n-1}{2}\in\mathbb{Z}^+\wedge m\in\mathbb{Z}^+\wedge m<n-1\wedge\frac{a}{b}\ngtr 0$, let $\frac{r}{s}=\left(-\frac{a}{b}\right)^{1/n}$, then

$$\int \frac{x^m}{a+b\,x^n}\,dx \to \frac{r^{m+1}}{a\,n\,s^m}\int \frac{1}{r-s\,x}\,dx-\frac{2\,(-r)^{m+1}}{a\,n\,s^m}\sum_{k=1}^{\frac{n-1}{2}}\int \frac{r\,\mathrm{Cos}\left[\frac{(2\,k-1)\,m\,\pi}{n}\right]+s\,\mathrm{Cos}\left[\frac{(2\,k-1)\,(m+1)\,\pi}{n}\right]\,x}{r^2+2\,r\,s\,\mathrm{Cos}\left[\frac{(2\,k-1)\,\pi}{n}\right]\,x+s^2\,x^2}\,dx$$

Program code:

```
Int[x_^m_./(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,n]], s=Denominator[Rt[-a/b,n]], k, u},
  u=Int[(r*Cos[(2*k-1)*m*Pi/n]+s*Cos[(2*k-1)*(m+1)*Pi/n]*x)/(r^2+2*r*s*Cos[(2*k-1)*Pi/n]*x+s^2*x^2),x];
  r^(m+1)/(a*n*s^m)*Int[1/(r-s*x),x] - Dist[2*(-r)^(m+1)/(a*n*s^m),Sum[u,{k,1,(n-1)/2}],x]] /;
FreeQ[{a,b},x] && IGtQ[(n-1)/2,0] && IGtQ[m,0] && LtQ[m,n-1] && NegQ[a/b]
```

2. $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n}{2}\in\mathbb{Z}^+\land m\in\mathbb{Z}^+\land m<n-1$

1. $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n-2}{4}\in\mathbb{Z}^+\land m\in\mathbb{Z}^+\land m<n-1$

1: $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n-2}{4}\in\mathbb{Z}^+\land m\in\mathbb{Z}^+\land m<n-1\land \frac{a}{b}>0$

Derivation: Algebraic expansion

Basis: If $\frac{n-2}{4}\in\mathbb{Z}^+\land m\in\mathbb{Z}^+\land m<n-1$, let $\frac{r}{s}=\left(\frac{a}{b}\right)^{1/n}$, then

$$\frac{z^m}{a+b\,z^n}==\frac{2\,(-1)^{\frac{m}{2}}\,r^{m+2}}{a\,n\,s^m\,(r^2+s^2\,z^2)}+\frac{4\,r^{m+2}}{a\,n\,s^m}\sum_{k=1}^{\frac{n-2}{4}}\frac{r^2\,\text{Cos}\left[\frac{(2\,k-1)\,m\,\pi}{n}\right]-s^2\,\text{Cos}\left[\frac{(2\,k-1)\,(m+2)\,\pi}{n}\right]\,z^2}{r^4-2\,r^2\,s^2\,\text{Cos}\left[\frac{2\,(2\,k-1)\,\pi}{n}\right]\,z^2+s^4\,z^4}$$

Basis: $\frac{r^2\,\text{Cos}[\rho]-s^2\,\text{Cos}[\rho+2\,\theta]\,z^2}{r^4-2\,r^2\,s^2\,\text{Cos}[2\,\theta]\,z^2+s^4\,z^4}==\frac{1}{2\,r}\left(\frac{r\,\text{Cos}[\rho]-s\,\text{Cos}[\rho+\theta]\,z}{r^2-2\,r\,s\,\text{Cos}[\theta]\,z+s^2\,z^2}+\frac{r\,\text{Cos}[\rho]+s\,\text{Cos}[\rho+\theta]\,z}{r^2+2\,r\,s\,\text{Cos}[\theta]\,z+s^2\,z^2}\right)$

Rule 1.1.3.2.7.1.4.1.2.1.1: If $\frac{n-2}{4}\in\mathbb{Z}^+\land m\in\mathbb{Z}^+\land m<n-1\land\frac{a}{b}>0$, let $\frac{r}{s}=\left(\frac{a}{b}\right)^{1/n}$, then

$$\int\frac{x^m}{a+b\,x^n}\,dx\to\frac{2\,(-1)^{\frac{m}{2}}\,r^{m+2}}{a\,n\,s^m}\int\frac{1}{r^2+s^2\,x^2}\,dx+\frac{4\,r^{m+2}}{a\,n\,s^m}\sum_{k=1}^{\frac{n-2}{4}}\int\frac{r^2\,\text{Cos}\left[\frac{(2\,k-1)\,m\,\pi}{n}\right]-s^2\,\text{Cos}\left[\frac{(2\,k-1)\,(m+2)\,\pi}{n}\right]\,x^2}{r^4-2\,r^2\,s^2\,\text{Cos}\left[\frac{2\,(2\,k-1)\,\pi}{n}\right]\,x^2+s^4\,x^4}\,dx$$

$$\to\frac{2\,(-1)^{\frac{m}{2}}\,r^{m+2}}{a\,n\,s^m}\int\frac{1}{r^2+s^2\,x^2}\,dx+\frac{2\,r^{m+1}}{a\,n\,s^m}\sum_{k=1}^{\frac{n-2}{4}}\left(\int\frac{r\,\text{Cos}\left[\frac{(2\,k-1)\,m\,\pi}{n}\right]-s\,\text{Cos}\left[\frac{(2\,k-1)\,(m+1)\,\pi}{n}\right]\,x}{r^2-2\,r\,s\,\text{Cos}\left[\frac{(2\,k-1)\,\pi}{n}\right]\,x+s^2\,x^2}\,dx+\int\frac{r\,\text{Cos}\left[\frac{(2\,k-1)\,m\,\pi}{n}\right]+s\,\text{Cos}\left[\frac{(2\,k-1)\,(m+1)\,\pi}{n}\right]\,x}{r^2+2\,r\,s\,\text{Cos}\left[\frac{(2\,k-1)\,\pi}{n}\right]\,x+s^2\,x^2}\,dx\right)$$

Program code:

```
Int[x_^m_./(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[a/b,n]], s=Denominator[Rt[a/b,n]], k, u},
  u=Int[(r*Cos[(2*k-1)*m*Pi/n]-s*Cos[(2*k-1)*(m+1)*Pi/n]*x)/(r^2-2*r*s*Cos[(2*k-1)*Pi/n]*x+s^2*x^2),x] +
    Int[(r*Cos[(2*k-1)*m*Pi/n]+s*Cos[(2*k-1)*(m+1)*Pi/n]*x)/(r^2+2*r*s*Cos[(2*k-1)*Pi/n]*x+s^2*x^2),x];
  2*(-1)^(m/2)*r^(m+2)/(a*n*s^m)*Int[1/(r^2+s^2*x^2),x] + Dist[2*r^(m+1)/(a*n*s^m),Sum[u,{k,1,(n-2)/4}],x]] /;
 FreeQ[{a,b},x] && IGtQ[(n-2)/4,0] && IGtQ[m,0] && LtQ[m,n-1] && PosQ[a/b]
```

2: $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n-2}{4} \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+ \land m < n-1 \land \frac{a}{b} \ngtr 0$

Derivation: Algebraic expansion

Basis: If $\frac{n-2}{4} \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+ \land m < n-1$, let $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/n}$, then

$$\frac{z^m}{a+b\,z^n} == \frac{2\,r^{m+2}}{a\,n\,s^m\,(r^2-s^2\,z^2)} + \frac{4\,r^{m+2}}{a\,n\,s^m}\sum_{k=1}^{\frac{n-2}{4}} \frac{r^2\,\text{Cos}\left[\frac{2\,k\,m\,\pi}{n}\right]-s^2\,\text{Cos}\left[\frac{2\,k\,(m+2)\,\pi}{n}\right]\,z^2}{r^4-2\,r^2\,s^2\,\text{Cos}\left[\frac{4\,k\,\pi}{n}\right]\,z^2+s^4\,z^4}$$

Basis: $\frac{r^2\,\text{Cos}[\rho]-s^2\,\text{Cos}[\rho+2\,\theta]\,z^2}{r^4-2\,r^2\,s^2\,\text{Cos}[2\,\theta]\,z^2+s^4\,z^4} == \frac{1}{2\,r}\left(\frac{r\,\text{Cos}[\rho]-s\,\text{Cos}[\rho+\theta]\,z}{r^2-2\,r\,s\,\text{Cos}[\theta]\,z+s^2\,z^2} + \frac{r\,\text{Cos}[\rho]+s\,\text{Cos}[\rho+\theta]\,z}{r^2+2\,r\,s\,\text{Cos}[\theta]\,z+s^2\,z^2}\right)$

Rule 1.1.3.2.7.1.4.1.2.1.2: If $\frac{n-2}{4} \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+ \land m < n-1 \land \frac{a}{b} \ngtr 0$, let $\frac{r}{s} = \left(-\frac{a}{b}\right)^{1/n}$, then

$$\int \frac{x^m}{a+b\,x^n}\,dx \to \frac{2\,r^{m+2}}{a\,n\,s^m}\int \frac{1}{r^2-s^2\,x^2}\,dx + \frac{4\,r^{m+2}}{a\,n\,s^m}\sum_{k=1}^{\frac{n-2}{4}}\int \frac{r^2\,\text{Cos}\left[\frac{2\,k\,m\,\pi}{n}\right]-s^2\,\text{Cos}\left[\frac{2\,k\,(m+2)\,\pi}{n}\right]\,x^2}{r^4-2\,r^2\,s^2\,\text{Cos}\left[\frac{4\,k\,\pi}{n}\right]\,x^2+s^4\,x^4}\,dx$$

$$\to \frac{2\,r^{m+2}}{a\,n\,s^m}\int \frac{1}{r^2-s^2\,x^2}\,dx + \frac{2\,r^{m+1}}{a\,n\,s^m}\sum_{k=1}^{\frac{n-2}{4}}\left(\int \frac{r\,\text{Cos}\left[\frac{2\,k\,m\,\pi}{n}\right]-s\,\text{Cos}\left[\frac{2\,k\,(m+1)\,\pi}{n}\right]\,x}{r^2-2\,r\,s\,\text{Cos}\left[\frac{2\,k\,\pi}{n}\right]\,x+s^2\,x^2}\,dx + \int \frac{r\,\text{Cos}\left[\frac{2\,k\,m\,\pi}{n}\right]+s\,\text{Cos}\left[\frac{2\,k\,(m+1)\,\pi}{n}\right]\,x}{r^2+2\,r\,s\,\text{Cos}\left[\frac{2\,k\,\pi}{n}\right]\,x+s^2\,x^2}\,dx\right)$$

Program code:

```
Int[x_^m_./(a_+b_.*x_^n_),x_Symbol] :=
  Module[{r=Numerator[Rt[-a/b,n]], s=Denominator[Rt[-a/b,n]], k, u},
  u=Int[(r*Cos[2*k*m*Pi/n]-s*Cos[2*k*(m+1)*Pi/n]*x)/(r^2-2*r*s*Cos[2*k*Pi/n]*x+s^2*x^2),x] +
    Int[(r*Cos[2*k*m*Pi/n]+s*Cos[2*k*(m+1)*Pi/n]*x)/(r^2+2*r*s*Cos[2*k*Pi/n]*x+s^2*x^2),x];
  2*r^(m+2)/(a*n*s^m)*Int[1/(r^2-s^2*x^2),x] + Dist[2*r^(m+1)/(a*n*s^m),Sum[u,{k,1,(n-2)/4}],x]] /;
 FreeQ[{a,b},x] && IGtQ[(n-2)/4,0] && IGtQ[m,0] && LtQ[m,n-1] && NegQ[a/b]
```

2. $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n}{4}\in\mathbb{Z}^+\wedge m\in\mathbb{Z}^+\wedge m<n-1$

1. $\int \frac{x^2}{a+b\,x^4}\,dx$

1: $\int \frac{x^2}{a+b\,x^4}\,dx$ when $\frac{a}{b}>0$

Derivation: Algebraic expansion

Basis: If $\frac{r}{s}=\sqrt{\frac{a}{b}}$, then $\frac{x^2}{a+b\,x^4}==\frac{r+s\,x^2}{2\,s\,(a+b\,x^4)}-\frac{r-s\,x^2}{2\,s\,(a+b\,x^4)}$

Note: Resulting integrands are of the form $\frac{d+e\,x^2}{a+c\,x^4}$ where $c\,d^2-a\,e^2==0$ as required by the algebraic trinomial rules.

Rule 1.1.3.2.7.1.4.1.2.2.1.1: If $\frac{a}{b}>0$, let $\frac{r}{s}=\sqrt{\frac{a}{b}}$, then

$$\int \frac{x^2}{a+b\,x^4}\,dx\to\frac{1}{2\,s}\int\frac{r+s\,x^2}{a+b\,x^4}\,dx-\frac{1}{2\,s}\int\frac{r-s\,x^2}{a+b\,x^4}\,dx$$

Program code:

```
Int[x_^2/(a_+b_.*x_^4),x_Symbol] :=
  With[{r=Numerator[Rt[a/b,2]], s=Denominator[Rt[a/b,2]]},
  1/(2*s)*Int[(r+s*x^2)/(a+b*x^4),x] -
  1/(2*s)*Int[(r-s*x^2)/(a+b*x^4),x]] /;
FreeQ[{a,b},x] && (GtQ[a/b,0] || PosQ[a/b] && AtomQ[SplitProduct[SumBaseQ,a]] && AtomQ[SplitProduct[SumBaseQ,b]])
```

2: $\int \frac{x^2}{a+b\,x^4}\,dx$ when $\frac{a}{b} \not> 0$

Reference: G&R 2.132.3.2', CRC 82'

Derivation: Algebraic expansion

Basis: If $\frac{r}{s} = \sqrt{-\frac{a}{b}}$, then $\frac{z}{a+b\,z^2} == \frac{s}{2\,b\,(r+s\,z)} - \frac{s}{2\,b\,(r-s\,z)}$

– Rule 1.1.3.2.7.1.4.1.2.2.1.2: If $\frac{a}{b} \not> 0$, let $\frac{r}{s} = \sqrt{-\frac{a}{b}}$, then

$$\int \frac{x^2}{a+b\,x^4}\,dx \to \frac{s}{2\,b} \int \frac{1}{r+s\,x^2}\,dx - \frac{s}{2\,b} \int \frac{1}{r-s\,x^2}\,dx$$

Program code:

```
Int[x_^2/(a_+b_.*x_^4),x_Symbol] :=
  With[{r=Numerator[Rt[-a/b,2]], s=Denominator[Rt[-a/b,2]]},
  s/(2*b)*Int[1/(r+s*x^2),x] -
  s/(2*b)*Int[1/(r-s*x^2),x]] /;
FreeQ[{a,b},x] && Not[GtQ[a/b,0]]
```

2. $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n}{4}\in\mathbb{Z}^+\land m\in\mathbb{Z}^+\land m<n-1\land n>4$

1: $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n}{4}\in\mathbb{Z}^+\land m\in\mathbb{Z}^+\land m<n-1\land \frac{a}{b}>0$

Reference: G&R 2.132.3.1', CRC 81'

Derivation: Algebraic expansion

Basis: If $\frac{r}{s}=\left(\frac{a}{b}\right)^{1/4}$, then $\frac{z}{a+b\,z^4}==\frac{s^3}{2\sqrt{2}\,b\,r\,\left(r^2-\sqrt{2}\,r\,s\,z+s^2\,z^2\right)}-\frac{s^3}{2\sqrt{2}\,b\,r\,\left(r^2+\sqrt{2}\,r\,s\,z+s^2\,z^2\right)}$

Rule 1.1.3.2.7.1.4.1.2.2.2.1: If $\frac{n}{4}\in\mathbb{Z}^+\land\ m\in\mathbb{Z}^+\land\ m<n-1\land\ \frac{a}{b}>0$, let $\frac{r}{s}=\left(\frac{a}{b}\right)^{1/4}$, then

$$\int \frac{x^m}{a+b\,x^n}\,dx\ \longrightarrow\ \frac{s^3}{2\sqrt{2}\,b\,r}\int \frac{x^{m-n/4}}{r^2-\sqrt{2}\,r\,s\,x^{n/4}+s^2\,x^{n/2}}\,dx-\frac{s^3}{2\sqrt{2}\,b\,r}\int \frac{x^{m-n/4}}{r^2+\sqrt{2}\,r\,s\,x^{n/4}+s^2\,x^{n/2}}\,dx$$

Program code:

```
Int[x_^m_./(a_+b_.*x_^n_),x_Symbol] :=
  With[{r=Numerator[Rt[a/b,4]], s=Denominator[Rt[a/b,4]]},
  s^3/(2*Sqrt[2]*b*r)*Int[x^(m-n/4)/(r^2-Sqrt[2]*r*s*x^(n/4)+s^2*x^(n/2)),x] -
  s^3/(2*Sqrt[2]*b*r)*Int[x^(m-n/4)/(r^2+Sqrt[2]*r*s*x^(n/4)+s^2*x^(n/2)),x]] /;
FreeQ[{a,b},x] && IGtQ[n/4,0] && IGtQ[m,0] && LtQ[m,n-1] && GtQ[a/b,0]
```

2. $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n}{4}\in\mathbb{Z}^+\land m\in\mathbb{Z}^+\land m<n-1\land\frac{a}{b}\not>0$

1: $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n}{4}\in\mathbb{Z}^+\land m\in\mathbb{Z}^+\land m<\frac{n}{2}\land\frac{a}{b}\not>0$

Reference: G&R 2.132.1.2', CRC 78'

Derivation: Algebraic expansion

Basis: If $\frac{r}{s}=\sqrt{-\frac{a}{b}}$, then $\frac{1}{a+b\,z^2}==\frac{r}{2\,a\,(r+s\,z)}+\frac{r}{2\,a\,(r-s\,z)}$

Rule 1.1.3.2.7.1.4.1.2.2.2.2.1: If $\frac{n}{4}\in\mathbb{Z}^+\land m\in\mathbb{Z}^+\land m<\frac{n}{2}\land\frac{a}{b}\not>0$, let $\frac{r}{s}=\sqrt{-\frac{a}{b}}$, then

$$\int \frac{x^m}{a+b\,x^n}\,dx\to\frac{r}{2\,a}\int\frac{x^m}{r+s\,x^{n/2}}\,dx+\frac{r}{2\,a}\int\frac{x^m}{r-s\,x^{n/2}}\,dx$$

Program code:

```
Int[x_^m_/(a_+b_.*x_^n_),x_Symbol] :=
  With[{r=Numerator[Rt[-a/b,2]], s=Denominator[Rt[-a/b,2]]},
  r/(2*a)*Int[x^m/(r+s*x^(n/2)),x] +
  r/(2*a)*Int[x^m/(r-s*x^(n/2)),x]] /;
FreeQ[{a,b},x] && IGtQ[n/4,0] && IGtQ[m,0] && LtQ[m,n/2] && Not[GtQ[a/b,0]]
```

2: $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{n}{4}\in\mathbb{Z}^+ \land m\in\mathbb{Z}^+ \land \frac{n}{2}\le m<n \land \frac{a}{b}\ngtr 0$

Reference: G&R 2.132.3.2', CRC 82'

Derivation: Algebraic expansion

Basis: If $\frac{r}{s}=\sqrt{-\frac{a}{b}}$, then $\frac{z}{a+b\,z^2}==\frac{s}{2\,b\,(r+s\,z)}-\frac{s}{2\,b\,(r-s\,z)}$

Rule 1.1.3.2.7.1.4.1.2.2.2.2.2: If $\frac{n}{4}\in\mathbb{Z}^+ \land m\in\mathbb{Z}^+ \land \frac{n}{2}\le m<n \land \frac{a}{b}\ngtr 0$, let $\frac{r}{s}=\sqrt{-\frac{a}{b}}$, then

$$\int \frac{x^m}{a+b\,x^n}\,dx \to \frac{s}{2\,b}\int \frac{x^{m-n/2}}{r+s\,x^{n/2}}\,dx-\frac{s}{2\,b}\int \frac{x^{m-n/2}}{r-s\,x^{n/2}}\,dx$$

Program code:

```
Int[x_^m_/(a_+b_.*x_^n_),x_Symbol] :=
  With[{r=Numerator[Rt[-a/b,2]], s=Denominator[Rt[-a/b,2]]},
  s/(2*b)*Int[x^(m-n/2)/(r+s*x^(n/2)),x] -
  s/(2*b)*Int[x^(m-n/2)/(r-s*x^(n/2)),x]] /;
FreeQ[{a,b},x] && IGtQ[n/4,0] && IGtQ[m,0] && LeQ[n/2,m] && LtQ[m,n] && Not[GtQ[a/b,0]]
```

2: $\int \frac{x^m}{a+b\,x^n}\,dx$ when $n \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+ \land m > 2\,n-1$

Derivation: Algebraic expansion

Rule 1.1.3.2.7.1.4.2: If $n \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+ \land m > 2\,n-1$, then

$$\int \frac{x^m}{a+b\,x^n}\,dx \longrightarrow \int \text{PolynomialDivide}\left[x^m,\ a+b\,x^n,\ x\right]\,dx$$

Program code:

```
Int[x_^m_/(a_+b_.*x_^n_),x_Symbol] :=
  Int[PolynomialDivide[x^m,(a+b*x^n),x],x] /;
FreeQ[{a,b},x] && IGtQ[m,0] && IGtQ[n,0] && GtQ[m,2*n-1]
```

5. $\int \frac{x^m}{\sqrt{a+b\,x^n}}\,dx$ when $n \in \mathbb{Z}^+ \land m \in \mathbb{Z}^+$

1. $\int \frac{x}{\sqrt{a+b\,x^3}}\,dx$

1: $\int \frac{x}{\sqrt{a+b\,x^3}}\,dx$ when $a > 0$

Derivation: Algebraic expansion

Note: $\frac{\sqrt{2}}{\sqrt{2+\sqrt{3}}} == -1+\sqrt{3}$

Rule: If $a > 0$, let $\frac{r}{s} \to \left(\frac{b}{a}\right)^{1/3}$, then

$$\int \frac{x}{\sqrt{a+b\,x^3}}\,dx \to \frac{\sqrt{2}\,s}{\sqrt{2+\sqrt{3}}\,r}\int \frac{1}{\sqrt{a+b\,x^3}}\,dx + \frac{1}{r}\int \frac{\left(1-\sqrt{3}\right)s+r\,x}{\sqrt{a+b\,x^3}}\,dx$$

Program code:

```
Int[x_/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{r=Numer[Rt[b/a,3]], s=Denom[Rt[b/a,3]]},
  Sqrt[2]*s/(Sqrt[2+Sqrt[3]]*r)*Int[1/Sqrt[a+b*x^3],x] + 1/r*Int[((1-Sqrt[3])*s+r*x)/Sqrt[a+b*x^3],x]] /;
FreeQ[{a,b},x] && PosQ[a]
```

2: $\int \frac{x}{\sqrt{a+b\,x^3}}\,dx$ when $a \ngtr 0$

Derivation: Algebraic expansion

Note: $\frac{\sqrt{2}}{\sqrt{2-\sqrt{3}}} == 1+\sqrt{3}$

Rule: If $a \ngtr 0$, let $\frac{r}{s} \to \left(\frac{b}{a}\right)^{1/3}$, then

$$\int \frac{x}{\sqrt{a+b\,x^3}}\,dx \to -\frac{\sqrt{2}\,s}{\sqrt{2-\sqrt{3}}\,r}\int \frac{1}{\sqrt{a+b\,x^3}}\,dx+\frac{1}{r}\int \frac{\left(1+\sqrt{3}\right)s+r\,x}{\sqrt{a+b\,x^3}}\,dx$$

Program code:

```
Int[x_/Sqrt[a_+b_.*x_^3],x_Symbol] :=
  With[{r=Numer[Rt[b/a,3]], s=Denom[Rt[b/a,3]]},
  -Sqrt[2]*s/(Sqrt[2-Sqrt[3]]*r)*Int[1/Sqrt[a+b*x^3],x] + 1/r*Int[((1+Sqrt[3])*s+r*x)/Sqrt[a+b*x^3],x]] /;
FreeQ[{a,b},x] && NegQ[a]
```

2. $\int \frac{x^2}{\sqrt{a+b x^4}} \, dx$

1: $\int \frac{x^2}{\sqrt{a+b x^4}} \, dx$ when $\frac{b}{a} > 0$

Derivation: Algebraic expansion

Rule 1.1.3.2.7.1.5.2.1: If $\frac{b}{a} > 0$, let $q \to \sqrt{\frac{b}{a}}$, then

$$\int \frac{x^2}{\sqrt{a+b x^4}} \, dx \longrightarrow \frac{1}{q} \int \frac{1}{\sqrt{a+b x^4}} \, dx - \frac{1}{q} \int \frac{1-q x^2}{\sqrt{a+b x^4}} \, dx$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  With[{q=Rt[b/a,2]},
  1/q*Int[1/Sqrt[a+b*x^4],x] - 1/q*Int[(1-q*x^2)/Sqrt[a+b*x^4],x]] /;
FreeQ[{a,b},x] && PosQ[b/a]
```

2. $\int \frac{x^2}{\sqrt{a+b x^4}} dx$ when $\frac{b}{a} \ngtr 0$

1: $\int \frac{x^2}{\sqrt{a+b x^4}} dx$ when $a < 0 \land b > 0$

Derivation: Algebraic expansion

- Rule 1.1.3.2.7.1.5.2.2.1: If $\frac{b}{a} \ngtr 0 \land a < 0$, let $q \to \sqrt{-\frac{b}{a}}$, then

$$\int \frac{x^2}{\sqrt{a+b x^4}} dx \longrightarrow \frac{1}{q} \int \frac{1}{\sqrt{a+b x^4}} dx - \frac{1}{q} \int \frac{1-q x^2}{\sqrt{a+b x^4}} dx$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  With[{q=Rt[-b/a,2]},
  1/q*Int[1/Sqrt[a+b*x^4],x] - 1/q*Int[(1-q*x^2)/Sqrt[a+b*x^4],x]] /;
FreeQ[{a,b},x] && LtQ[a,0] && GtQ[b,0]
```

2: $\int \frac{x^2}{\sqrt{a+b x^4}}\,dx$ when $\frac{b}{a} \ngtr 0 \land a \nless 0$

Derivation: Algebraic expansion

Rule 1.1.3.2.7.1.5.2.2.2: If $\frac{b}{a} \ngtr 0 \land a \nless 0$, let $q \to \sqrt{-\frac{b}{a}}$, then

$$\int \frac{x^2}{\sqrt{a+b x^4}}\,dx \to -\frac{1}{q}\int \frac{1}{\sqrt{a+b x^4}}\,dx + \frac{1}{q}\int \frac{1+q x^2}{\sqrt{a+b x^4}}\,dx$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^4],x_Symbol] :=
  With[{q=Rt[-b/a,2]},
  -1/q*Int[1/Sqrt[a+b*x^4],x] + 1/q*Int[(1+q*x^2)/Sqrt[a+b*x^4],x]] /;
FreeQ[{a,b},x] && NegQ[b/a]
```

3: $\int \frac{x^4}{\sqrt{a+b x^6}}\,dx$

Derivation: Algebraic expansion

Rule 1.1.3.2.7.1.5.3: Let $\frac{r}{s} \to \left(\frac{b}{a}\right)^{1/3}$, then

$$\int \frac{x^4}{\sqrt{a+b x^6}}\,dx \to \frac{\left(\sqrt{3}-1\right) s^2}{2 r^2}\int \frac{1}{\sqrt{a+b x^6}}\,dx - \frac{1}{2 r^2}\int \frac{\left(\sqrt{3}-1\right) s^2 - 2 r^2 x^4}{\sqrt{a+b x^6}}\,dx$$

$$\to \frac{\left(1+\sqrt{3}\right) r x \sqrt{a+b x^6}}{2 b \left(s+\left(1+\sqrt{3}\right) r x^2\right)} - \frac{3^{1/4} s x \left(s+r x^2\right) \sqrt{\frac{s^2-r s x^2+r^2 x^4}{\left(s+\left(1+\sqrt{3}\right) r x^2\right)^2}}}{2 r^2 \sqrt{a+b x^6} \sqrt{\frac{r x^2 \left(s+r x^2\right)}{\left(s+\left(1+\sqrt{3}\right) r x^2\right)^2}}} \text{EllipticE}\left[\text{ArcCos}\left[\frac{s+\left(1-\sqrt{3}\right) r x^2}{s+\left(1+\sqrt{3}\right) r x^2}\right], \frac{2+\sqrt{3}}{4}\right] -$$

$$\frac{\left(1-\sqrt{3}\right) s\, x\, \left(s+r x^2\right) \sqrt{\frac{s^2-r s x^2+r^2 x^4}{\left(s+\left(1+\sqrt{3}\right) r x^2\right)^2}}}{4\times 3^{1/4} r^2 \sqrt{a+b x^6} \sqrt{\frac{r x^2 \left(s+r x^2\right)}{\left(s+\left(1+\sqrt{3}\right) r x^2\right)^2}}} \text{EllipticF}\left[\text{ArcCos}\left[\frac{s+\left(1-\sqrt{3}\right) r x^2}{s+\left(1+\sqrt{3}\right) r x^2}\right], \frac{2+\sqrt{3}}{4}\right]$$

Program code:

```
Int[x_^4/Sqrt[a_+b_.*x_^6],x_Symbol] :=
  With[{r=Numer[Rt[b/a,3]], s=Denom[Rt[b/a,3]]},
  (Sqrt[3]-1)*s^2/(2*r^2)*Int[1/Sqrt[a+b*x^6],x] - 1/(2*r^2)*Int[((Sqrt[3]-1)*s^2-2*r^2*x^4)/Sqrt[a+b*x^6],x]] /;
FreeQ[{a,b},x]
```

```
(* Int[x_^4/Sqrt[a_+b_.*x_^6],x_Symbol] :=
  With[{r=Numer[Rt[b/a,3]], s=Denom[Rt[b/a,3]]},
  (1+Sqrt[3])*r*x*Sqrt[a+b*x^6]/(2*b*(s+(1+Sqrt[3])*r*x^2)) -
  3^(1/4)*s*x*(s+r*x^2)*Sqrt[(s^2-r*s*x^2+r^2*x^4)/(s+(1+Sqrt[3])*r*x^2)^2]/
    (2*r^2*Sqrt[a+b*x^6]*Sqrt[r*x^2*(s+r*x^2)/(s+(1+Sqrt[3])*r*x^2)^2])*
    EllipticE[ArcCos[(s+(1-Sqrt[3])*r*x^2)/(s+(1+Sqrt[3])*r*x^2)],(2+Sqrt[3])/4] -
  (1-Sqrt[3])*s*x*(s+r*x^2)*Sqrt[(s^2-r*s*x^2+r^2*x^4)/(s+(1+Sqrt[3])*r*x^2)^2]/
    (4*3^(1/4)*r^2*Sqrt[a+b*x^6]*Sqrt[r*x^2*(s+r*x^2)/(s+(1+Sqrt[3])*r*x^2)^2])*
    EllipticF[ArcCos[(s+(1-Sqrt[3])*r*x^2)/(s+(1+Sqrt[3])*r*x^2)],(2+Sqrt[3])/4]] /;
FreeQ[{a,b},x] *)
```

4: $\int \frac{x^2}{\sqrt{a+b x^8}}\,dx$

Derivation: Algebraic expansion

Basis: $\frac{x^2}{\sqrt{a+b x^8}} == \frac{1+\left(\frac{b}{a}\right)^{1/4} x^2}{2\left(\frac{b}{a}\right)^{1/4}\sqrt{a+b x^8}} - \frac{1-\left(\frac{b}{a}\right)^{1/4} x^2}{2\left(\frac{b}{a}\right)^{1/4}\sqrt{a+b x^8}}$

Note: Integrands are of the form $\frac{c+d x^2}{\sqrt{a+b x^8}}$ where $b c^4 - a d^4 == 0$ for which there is a terminal rule.

Rule 1.1.3.2.7.1.5.4:

$$\int \frac{x^2}{\sqrt{a+b x^8}}\,dx \longrightarrow \frac{1}{2\left(\frac{b}{a}\right)^{1/4}}\int \frac{1+\left(\frac{b}{a}\right)^{1/4} x^2}{\sqrt{a+b x^8}}\,dx - \frac{1}{2\left(\frac{b}{a}\right)^{1/4}}\int \frac{1-\left(\frac{b}{a}\right)^{1/4} x^2}{\sqrt{a+b x^8}}\,dx$$

Program code:

```
Int[x_^2/Sqrt[a_+b_.*x_^8],x_Symbol] :=
  1/(2*Rt[b/a,4])*Int[(1+Rt[b/a,4]*x^2)/Sqrt[a+b*x^8],x] -
  1/(2*Rt[b/a,4])*Int[(1-Rt[b/a,4]*x^2)/Sqrt[a+b*x^8],x] /;
FreeQ[{a,b},x]
```

6. $\int \frac{x^m}{(a+b\,x^n)^{1/4}}\,dx$ when $n \in \mathbb{Z}^+ \land 2\,m \in \mathbb{Z}^+$

1. $\int \frac{x^2}{(a+b\,x^4)^{1/4}}\,dx$

1: $\int \frac{x^2}{(a+b\,x^4)^{1/4}}\,dx$ when $\frac{b}{a} > 0$

Reference: G&R 2.110.1, CRC 88b

Derivation: Binomial recurrence 1b

Rule 1.1.3.2.7.1.6.1.1: If $\frac{b}{a} > 0$, then

$$\int \frac{x^2}{(a+b\,x^4)^{1/4}}\,dx \longrightarrow \frac{x^3}{2\,(a+b\,x^4)^{1/4}} - \frac{a}{2}\int \frac{x^2}{(a+b\,x^4)^{5/4}}\,dx$$

Program code:

```
Int[x_^2/(a_+b_.*x_^4)^(1/4),x_Symbol] :=
  x^3/(2*(a+b*x^4)^(1/4)) - a/2*Int[x^2/(a+b*x^4)^(5/4),x] /;
FreeQ[{a,b},x] && PosQ[b/a]
```

2: $\int \frac{x^2}{(a+b\,x^4)^{1/4}}\,dx$ when $\frac{b}{a} \ngtr 0$

Reference: G&R 2.110.5, CRC 88a

Derivation: Binomial recurrence 3a

Rule 1.1.3.2.7.1.6.1.2: If $\frac{b}{a} \ngtr 0$, then

$$\int \frac{x^2}{\left(a+b\,x^4\right)^{1/4}}\,dx \longrightarrow \frac{\left(a+b\,x^4\right)^{3/4}}{2\,b\,x}+\frac{a}{2\,b}\int \frac{1}{x^2\,\left(a+b\,x^4\right)^{1/4}}\,dx$$

Program code:

```
Int[x_^2/(a_+b_.*x_^4)^(1/4),x_Symbol] :=
  (a+b*x^4)^(3/4)/(2*b*x) + a/(2*b)*Int[1/(x^2*(a+b*x^4)^(1/4)),x] /;
FreeQ[{a,b},x] && NegQ[b/a]
```

2. $\int \frac{1}{x^2\,\left(a+b\,x^4\right)^{1/4}}\,dx$

   1: $\int \frac{1}{x^2\,\left(a+b\,x^4\right)^{1/4}}\,dx$ when $\frac{b}{a}>0$

Reference: G&R 2.110.3

Derivation: Binomial recurrence 1a

Derivation: Integration by parts

Rule 1.1.3.2.7.1.6.2.1: If $\frac{b}{a}>0$, then

$$\int \frac{1}{x^2\,\left(a+b\,x^4\right)^{1/4}}\,dx \longrightarrow -\frac{1}{x\,\left(a+b\,x^4\right)^{1/4}}-b\int \frac{x^2}{\left(a+b\,x^4\right)^{5/4}}\,dx$$

Program code:

```
Int[1/(x_^2*(a_+b_.*x_^4)^(1/4)),x_Symbol] :=
  -1/(x*(a+b*x^4)^(1/4)) - b*Int[x^2/(a+b*x^4)^(5/4),x] /;
FreeQ[{a,b},x] && PosQ[b/a]
```

2: $\int \frac{1}{x^2 \left(a+b x^4\right)^{1/4}} \, dx$ when $\frac{b}{a} \ngtr 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x \left(1+\frac{a}{b x^4}\right)^{1/4}}{\left(a+b x^4\right)^{1/4}} == 0$

Rule 1.1.3.2.7.1.6.2.2: If $\frac{b}{a} \ngtr 0$, then

$$\int \frac{1}{x^2 \left(a+b x^4\right)^{1/4}} \, dx \longrightarrow \frac{x \left(1+\frac{a}{b x^4}\right)^{1/4}}{\left(a+b x^4\right)^{1/4}} \int \frac{1}{x^3 \left(1+\frac{a}{b x^4}\right)^{1/4}} \, dx$$

Program code:

```
Int[1/(x_^2*(a_+b_.*x_^4)^(1/4)),x_Symbol] :=
  x*(1+a/(b*x^4))^(1/4)/(a+b*x^4)^(1/4)*Int[1/(x^3*(1+a/(b*x^4))^(1/4)),x] /;
FreeQ[{a,b},x] && NegQ[b/a]
```

3. $\int \frac{\sqrt{c\,x}}{\left(a+b\,x^2\right)^{1/4}}\,dx$

1: $\int \frac{\sqrt{c\,x}}{\left(a+b\,x^2\right)^{1/4}}\,dx$ when $\frac{b}{a}>0$

Reference: G&R 2.110.1, CRC 88b

Derivation: Binomial recurrence 1b

Rule 1.1.3.2.7.1.6.3.2: If $\frac{b}{a}>0$, then

$$\int \frac{\sqrt{c\,x}}{\left(a+b\,x^2\right)^{1/4}}\,dx \to \frac{x\sqrt{c\,x}}{\left(a+b\,x^2\right)^{1/4}}-\frac{a}{2}\int \frac{\sqrt{c\,x}}{\left(a+b\,x^2\right)^{5/4}}\,dx$$

Program code:

```
Int[Sqrt[c_*x_]/(a_+b_.*x_^2)^(1/4),x_Symbol] :=
  x*Sqrt[c*x]/(a+b*x^2)^(1/4) - a/2*Int[Sqrt[c*x]/(a+b*x^2)^(5/4),x] /;
FreeQ[{a,b,c},x] && PosQ[b/a]
```

2: $\int \frac{\sqrt{c\,x}}{\left(a+b\,x^2\right)^{1/4}}\,dx$ when $\frac{b}{a} \ngtr 0$

Reference: G&R 2.110.5, CRC 88a

Derivation: Binomial recurrence 3a

Rule 1.1.3.2.7.1.6.3.2: If $\frac{b}{a} \ngtr 0$, then

$$\int \frac{\sqrt{c\,x}}{\left(a+b\,x^2\right)^{1/4}}\,dx \to \frac{c\left(a+b\,x^2\right)^{3/4}}{b\,\sqrt{c\,x}}+\frac{a\,c^2}{2\,b}\int \frac{1}{(c\,x)^{3/2}\left(a+b\,x^2\right)^{1/4}}\,dx$$

Program code:

```
Int[Sqrt[c_*x_]/(a_+b_.*x_^2)^(1/4),x_Symbol] :=
  c*(a+b*x^2)^(3/4)/(b*Sqrt[c*x]) + a*c^2/(2*b)*Int[1/((c*x)^(3/2)*(a+b*x^2)^(1/4)),x] /;
FreeQ[{a,b,c},x] && NegQ[b/a]
```

4. $\int \frac{1}{(c\,x)^{3/2}\left(a+b\,x^2\right)^{1/4}}\,dx$

1: $\int \frac{1}{(c\,x)^{3/2}\left(a+b\,x^2\right)^{1/4}}\,dx$ when $\frac{b}{a} > 0$

Reference: G&R 2.110.3

Derivation: Binomial recurrence 1a

Derivation: Integration by parts

Rule 1.1.3.2.7.1.6.4.1: If $\frac{b}{a} > 0$, then

$$\int \frac{1}{(c\,x)^{3/2}\,\left(a+b\,x^2\right)^{1/4}}\,dx \to -\frac{2}{c\,\sqrt{c\,x}\,\left(a+b\,x^2\right)^{1/4}}-\frac{b}{c^2}\int \frac{\sqrt{c\,x}}{\left(a+b\,x^2\right)^{5/4}}\,dx$$

Program code:

```
Int[1/((c_.*x_)^(3/2)*(a_+b_.*x_^2)^(1/4)),x_Symbol] :=
  -2/(c*Sqrt[c*x]*(a+b*x^2)^(1/4)) - b/c^2*Int[Sqrt[c*x]/(a+b*x^2)^(5/4),x] /;
FreeQ[{a,b,c},x] && PosQ[b/a]
```

2: $\int \frac{1}{(c\,x)^{3/2}\,\left(a+b\,x^2\right)^{1/4}}\,dx$ when $\frac{b}{a} \not> 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{c\,x}\,\left(1+\frac{a}{b\,x^2}\right)^{1/4}}{\left(a+b\,x^2\right)^{1/4}} == 0$

Rule 1.1.3.2.7.1.6.4.2: If $\frac{b}{a} \not> 0$, then

$$\int \frac{1}{(c\,x)^{3/2}\,\left(a+b\,x^2\right)^{1/4}}\,dx \to \frac{\sqrt{c\,x}\,\left(1+\frac{a}{b\,x^2}\right)^{1/4}}{c^2\,\left(a+b\,x^2\right)^{1/4}}\int \frac{1}{x^2\,\left(1+\frac{a}{b\,x^2}\right)^{1/4}}\,dx$$

Program code:

```
Int[1/((c_.*x_)^(3/2)*(a_+b_.*x_^2)^(1/4)),x_Symbol] :=
  Sqrt[c*x]*(1+a/(b*x^2))^(1/4)/(c^2*(a+b*x^2)^(1/4))*Int[1/(x^2*(1+a/(b*x^2))^(1/4)),x] /;
FreeQ[{a,b,c},x] && NegQ[b/a]
```

7. $\int \frac{\sqrt{c\,x}}{\sqrt{a+b\,x^2}}\,dx$ when $-\frac{b}{a} > 0$

1. $\int \frac{\sqrt{x}}{\sqrt{a+b x^2}} \, dx$ when $-\frac{b}{a} > 0$

1: $\int \frac{\sqrt{x}}{\sqrt{a+b x^2}} \, dx$ when $-\frac{b}{a} > 0 \land a > 0$

Derivation: Integration by substitution

Basis: If $-\frac{b}{a} > 0 \land a > 0$, then $\frac{\sqrt{x}}{\sqrt{a+b x^2}} == -\frac{2}{\sqrt{a}\left(-\frac{b}{a}\right)^{3/4}} \text{Subst}\left[\frac{\sqrt{1-2x^2}}{\sqrt{1-x^2}}, x, \frac{\sqrt{1-\sqrt{-\frac{b}{a}}\, x}}{\sqrt{2}}\right] \partial_x \frac{\sqrt{1-\sqrt{-\frac{b}{a}}\, x}}{\sqrt{2}}$

Rule 1.1.3.2.7.1.7.1.1: If $-\frac{b}{a} > 0 \land a > 0$, then

$$\int \frac{\sqrt{x}}{\sqrt{a+b x^2}} \, dx \to -\frac{2}{\sqrt{a}\left(-\frac{b}{a}\right)^{3/4}} \text{Subst}\left[\int \frac{\sqrt{1-2x^2}}{\sqrt{1-x^2}} \, dx, x, \frac{\sqrt{1-\sqrt{-\frac{b}{a}}\, x}}{\sqrt{2}}\right]$$

Program code:

```
Int[Sqrt[x_]/Sqrt[a_+b_.*x_^2],x_Symbol] :=
  -2/(Sqrt[a]*(-b/a)^(3/4))*Subst[Int[Sqrt[1-2*x^2]/Sqrt[1-x^2],x],x,Sqrt[1-Sqrt[-b/a]*x]/Sqrt[2]] /;
FreeQ[{a,b},x] && GtQ[-b/a,0] && GtQ[a,0]
```

2: $\int \frac{\sqrt{x}}{\sqrt{a+b x^2}} \, dx$ when $-\frac{b}{a} > 0 \land a \not> 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{1+\frac{b x^2}{a}}}{\sqrt{a+b x^2}} == 0$

Rule 1.1.3.2.7.1.7.1.2: If $-\frac{b}{a} > 0 \land a \not> 0$, then

$$\int \frac{\sqrt{x}}{\sqrt{a+b x^2}}\, dx \to \frac{\sqrt{1+\frac{b x^2}{a}}}{\sqrt{a+b x^2}} \int \frac{\sqrt{x}}{\sqrt{1+\frac{b x^2}{a}}}\, dx$$

Program code:

```
Int[Sqrt[x_]/Sqrt[a_+b_.*x_^2],x_Symbol] :=
  Sqrt[1+b*x^2/a]/Sqrt[a+b*x^2]*Int[Sqrt[x]/Sqrt[1+b*x^2/a],x] /;
FreeQ[{a,b},x] && GtQ[-b/a,0] && Not[GtQ[a,0]]
```

2: $\int \frac{\sqrt{c x}}{\sqrt{a+b x^2}}\, dx$ when $-\frac{b}{a} > 0$

Derivation: Piecewise constant extraction

Rule 1.1.3.2.7.1.7.2: If $-\frac{b}{a} > 0$, then

$$\int \frac{\sqrt{c x}}{\sqrt{a+b x^2}}\, dx \to \frac{\sqrt{c x}}{\sqrt{x}} \int \frac{\sqrt{x}}{\sqrt{a+b x^2}}\, dx$$

Program code:

```
Int[Sqrt[c_*x_]/Sqrt[a_+b_.*x_^2],x_Symbol] :=
  Sqrt[c*x]/Sqrt[x]*Int[Sqrt[x]/Sqrt[a+b*x^2],x] /;
FreeQ[{a,b,c},x] && GtQ[-b/a,0]
```

8: $\int (c\,x)^m\,\left(a+b\,x^n\right)^p\,dx$ when $n\in\mathbb{Z}^+\land\,m>n-1\,\land\,m+n\,p+1\neq 0$

Reference: G&R 2.110.5, CRC 88a

Derivation: Binomial recurrence 3a

Derivation: Inverted integration by parts

Rule 1.1.3.2.7.1.8: If $n\in\mathbb{Z}^+\land\,m>n-1\,\land\,m+n\,p+1\neq 0$, then

$$\int (c\,x)^m\,\left(a+b\,x^n\right)^p\,dx\,\to\,\frac{c^{n-1}\,(c\,x)^{m-n+1}\,\left(a+b\,x^n\right)^{p+1}}{b\,(m+n\,p+1)}-\frac{a\,c^n\,(m-n+1)}{b\,(m+n\,p+1)}\int (c\,x)^{m-n}\,\left(a+b\,x^n\right)^p\,dx$$

Program code:

```
Int[(c_.*x_)^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  c^(n-1)*(c*x)^(m-n+1)*(a+b*x^n)^(p+1)/(b*(m+n*p+1)) -
  a*c^n*(m-n+1)/(b*(m+n*p+1))*Int[(c*x)^(m-n)*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,p},x] && IGtQ[n,0] && GtQ[m,n-1] && NeQ[m+n*p+1,0] && IntBinomialQ[a,b,c,n,m,p,x]
```

```
Int[(c_.*x_)^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  c^(n-1)*(c*x)^(m-n+1)*(a+b*x^n)^(p+1)/(b*(m+n*p+1)) -
  a*c^n*(m-n+1)/(b*(m+n*p+1))*Int[(c*x)^(m-n)*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,m,p},x] && IGtQ[n,0] && SumSimplerQ[m,-n] && NeQ[m+n*p+1,0] && ILtQ[Simplify[(m+1)/n+p],0]
```

```
Int[(c_.*x_)^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  c^(2*n-1)*(c*x)^(m-2*n+1)*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(b1*b2*(m+2*n*p+1)) -
  a1*a2*c^(2*n)*(m-2*n+1)/(b1*b2*(m+2*n*p+1))*Int[(c*x)^(m-2*n)*(a1+b1*x^n)^p*(a2+b2*x^n)^p,x] /;
FreeQ[{a1,b1,a2,b2,c,p},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && GtQ[m,2*n-1] && NeQ[m+2*n*p+1,0] &&
  IntBinomialQ[a1*a2,b1*b2,c,2*n,m,p,x]
```

```
Int[(c_.*x_)^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  c^(2*n-1)*(c*x)^(m-2*n+1)*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(b1*b2*(m+2*n*p+1)) -
  a1*a2*c^(2*n)*(m-2*n+1)/(b1*b2*(m+2*n*p+1))*Int[(c*x)^(m-2*n)*(a1+b1*x^n)^p*(a2+b2*x^n)^p,x] /;
FreeQ[{a1,b1,a2,b2,c,m,p},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && SumSimplerQ[m,-2*n] && NeQ[m+2*n*p+1,0] &&
  ILtQ[Simplify[(m+1)/(2*n)+p],0]
```

9: $\int (c\,x)^m\,(a+b\,x^n)^p\,dx$ when $n\in\mathbb{Z}^+\wedge m<-1$

Reference: G&R 2.110.6, CRC 88c

Derivation: Binomial recurrence 3b

Derivation: Integration by parts

Basis: $x^m\,(a+b\,x^n)^p == \frac{x^m}{(a+b\,x^n)^{\frac{m+n+1}{n}}}\,(a+b\,x^n)^{\frac{m+1}{n}+p+1}$

Basis: $\int \frac{x^m}{(a+b\,x^n)^{\frac{m+n+1}{n}}}\,dx == \frac{x^{m+1}}{(a+b\,x^n)^{\frac{m+1}{n}}\,(a\,(m+1))}$

Note: Requirement that $m+1<n$ ensures new term is a proper fraction.

Rule 1.1.3.2.7.1.9: If $n\in\mathbb{Z}^+\wedge m<-1$, then

$$\int (c\,x)^m\,(a+b\,x^n)^p\,dx \longrightarrow \frac{(c\,x)^{m+1}\,(a+b\,x^n)^{p+1}}{a\,c\,(m+1)}-\frac{b\,(m+n\,(p+1)+1)}{a\,c^n\,(m+1)}\int (c\,x)^{m+n}\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[(c_.*x_)^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a+b*x^n)^(p+1)/(a*c*(m+1)) -
  b*(m+n*(p+1)+1)/(a*c^n*(m+1))*Int[(c*x)^(m+n)*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,p},x] && IGtQ[n,0] && LtQ[m,-1] && IntBinomialQ[a,b,c,n,m,p,x]
```

```
Int[(c_.*x_)^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a+b*x^n)^(p+1)/(a*c*(m+1)) -
  b*(m+n*(p+1)+1)/(a*c^n*(m+1))*Int[(c*x)^(m+n)*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,m,p},x] && IGtQ[n,0] && SumSimplerQ[m,n] && ILtQ[Simplify[(m+1)/n+p],0]
```

```
Int[(c_.*x_)^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(a1*a2*c*(m+1)) -
  b1*b2*(m+2*n*(p+1)+1)/(a1*a2*c^(2*n)*(m+1))*Int[(c*x)^(m+2*n)*(a1+b1*x^n)^p*(a2+b2*x^n)^p,x] /;
FreeQ[{a1,b1,a2,b2,c,p},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && LtQ[m,-1] && IntBinomialQ[a1*a2,b1*b2,c,2*n,m,p,x]
```

```
Int[(c_.*x_)^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(a1*a2*c*(m+1)) -
  b1*b2*(m+2*n*(p+1)+1)/(a1*a2*c^(2*n)*(m+1))*Int[(c*x)^(m+2*n)*(a1+b1*x^n)^p*(a2+b2*x^n)^p,x] /;
FreeQ[{a1,b1,a2,b2,c,m,p},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && SumSimplerQ[m,2*n] && ILtQ[Simplify[(m+1)/(2*n)+p],0]
```

10: $\int (c\,x)^m\,(a+b\,x^n)^p\,dx$ when $n \in \mathbb{Z}^+ \land m \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $(c\,x)^m\,F[x] = \frac{k}{c}\,\text{Subst}\left[x^{k\,(m+1)-1}\,F\left[\frac{x^k}{c}\right],\,x,\,(c\,x)^{1/k}\right]\,\partial_x\,(c\,x)^{1/k}$

Rule 1.1.3.2.7.1.10: If $n \in \mathbb{Z}^+ \land m \in \mathbb{F}$, let $k = \text{Denominator}[m]$, then

$$\int (c\,x)^m\,(a+b\,x^n)^p\,dx \;\longrightarrow\; \frac{k}{c}\,\text{Subst}\left[\int x^{k\,(m+1)-1}\left(a+\frac{b\,x^{k\,n}}{c^n}\right)^p dx,\,x,\,(c\,x)^{1/k}\right]$$

Program code:

```
Int[(c_.*x_)^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  k/c*Subst[Int[x^(k*(m+1)-1)*(a+b*x^(k*n)/c^n)^p,x],x,(c*x)^(1/k)]] /;
FreeQ[{a,b,c,p},x] && IGtQ[n,0] && FractionQ[m] && IntBinomialQ[a,b,c,n,m,p,x]
```

```
Int[(c_.*x_)^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  k/c*Subst[Int[x^(k*(m+1)-1)*(a1+b1*x^(k*n)/c^n)^p*(a2+b2*x^(k*n)/c^n)^p,x],x,(c*x)^(1/k)]] /;
FreeQ[{a1,b1,a2,b2,c,p},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && FractionQ[m] && IntBinomialQ[a1*a2,b1*b2,c,2*n,m,p,x]
```

11. $\int x^m \left(a+b\,x^n\right)^p dx$ when $n\in\mathbb{Z}^+\land -1<p<0\land p\neq -\frac{1}{2}\land m\in\mathbb{Z}\land \text{Denominator}\left[p+\frac{m+1}{n}\right]<\text{Denominator}[p]$

1: $\int x^m \left(a+b\,x^n\right)^p dx$ when $n\in\mathbb{Z}^+\land -1<p<0\land p\neq -\frac{1}{2}\land m\in\mathbb{Z}\land p+\frac{m+1}{n}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}\land m\in\mathbb{Z}\land p+\frac{m+1}{n}\in\mathbb{Z}$, then

$$x^m \left(a+b\,x^n\right)^p == a^{p+\frac{m+1}{n}}\,\text{Subst}\left[\frac{x^m}{\left(1-b\,x^n\right)^{p+\frac{m+1}{n}+1}},\,x,\,\frac{x}{\left(a+b\,x^n\right)^{1/n}}\right]\partial_x \frac{x}{\left(a+b\,x^n\right)^{1/n}}$$

Rule 1.1.3.2.7.1.11.1: If $n\in\mathbb{Z}^+\land -1<p<0\land p\neq -\frac{1}{2}\land m\in\mathbb{Z}\land p+\frac{m+1}{n}\in\mathbb{Z}$, then

$$\int x^m \left(a+b\,x^n\right)^p dx \,\longrightarrow\, a^{p+\frac{m+1}{n}}\,\text{Subst}\left[\int \frac{x^m}{\left(1-b\,x^n\right)^{p+\frac{m+1}{n}+1}}\,dx,\,x,\,\frac{x}{\left(a+b\,x^n\right)^{1/n}}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  a^(p+(m+1)/n)*Subst[Int[x^m/(1-b*x^n)^(p+(m+1)/n+1),x],x,x/(a+b*x^n)^(1/n)] /;
FreeQ[{a,b},x] && IGtQ[n,0] && LtQ[-1,p,0] && NeQ[p,-1/2] && IntegersQ[m,p+(m+1)/n]
```

```
Int[x_^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  (a1*a2)^(p+(m+1)/(2*n))*
    Subst[Int[x^m/((1-b1*x^n)^(p+(m+1)/(2*n)+1)*(1-b2*x^n)^(p+(m+1)/(2*n)+1)),x],x,
      x/((a1+b1*x^n)^(1/(2*n))*(a2+b2*x^n)^(1/(2*n)))] /;
FreeQ[{a1,b1,a2,b2},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && LtQ[-1,p,0] && NeQ[p,-1/2] && IntegersQ[m,p+(m+1)/(2*n)]
```

2: $\int x^m (a+b x^n)^p \, dx$ when $n \in \mathbb{Z}^+ \land -1 < p < 0 \land p \neq -\frac{1}{2} \land m \in \mathbb{Z} \land \text{Denominator}\left[p+\frac{m+1}{n}\right] < \text{Denominator}[p]$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \left(\left(\frac{a}{a+b x^n}\right)^{p+\frac{m+1}{n}} (a+b x^n)^{p+\frac{m+1}{n}} \right) == 0$

Basis: If $n \in \mathbb{Z}$, then $\frac{x^m}{\left(\frac{a}{a+b x^n}\right)^{p+\frac{m+1}{n}} (a+b x^n)^{p+\frac{m+1}{n}}} == \text{Subst}\left[\frac{x^m}{(1-b x^n)^{p+\frac{m+1}{n}+1}}, x, \frac{x}{(a+b x^n)^{1/n}}\right] \partial_x \frac{x}{(a+b x^n)^{1/n}}$

Rule 1.1.3.2.7.1.11.2: If $n \in \mathbb{Z}^+ \land -1 < p < 0 \land p \neq -\frac{1}{2} \land m \in \mathbb{Z} \land \text{Denominator}\left[p+\frac{m+1}{n}\right] < \text{Denominator}[p]$, then

$$\int x^m (a+b x^n)^p \, dx \rightarrow \left(\frac{a}{a+b x^n}\right)^{p+\frac{m+1}{n}} (a+b x^n)^{p+\frac{m+1}{n}} \int \frac{x^m}{\left(\frac{a}{a+b x^n}\right)^{p+\frac{m+1}{n}} (a+b x^n)^{\frac{m+1}{n}}} \, dx$$

$$\rightarrow \left(\frac{a}{a+b x^n}\right)^{p+\frac{m+1}{n}} (a+b x^n)^{p+\frac{m+1}{n}} \text{Subst}\left[\int \frac{x^m}{(1-b x^n)^{p+\frac{m+1}{n}+1}} \, dx, x, \frac{x}{(a+b x^n)^{1/n}}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  (a/(a+b*x^n))^(p+(m+1)/n)*(a+b*x^n)^(p+(m+1)/n)*Subst[Int[x^m/(1-b*x^n)^(p+(m+1)/n+1),x],x,x/(a+b*x^n)^(1/n)] /;
FreeQ[{a,b},x] && IGtQ[n,0] && LtQ[-1,p,0] && NeQ[p,-1/2] && IntegerQ[m] && LtQ[Denominator[p+(m+1)/n],Denominator[p]]
```

```
Int[x_^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  (a1/(a1+b1*x^n))^(p+(m+1)/(2*n))*(a1+b1*x^n)^(p+(m+1)/(2*n))*(a2/(a2+b2*x^n))^(p+(m+1)/(2*n))*(a2+b2*x^n)^(p+(m+1)/(2*n))*
    Subst[Int[x^m/((1-b1*x^n)^(p+(m+1)/(2*n)+1)*(1-b2*x^n)^(p+(m+1)/(2*n)+1)),x],x,
      x/((a1+b1*x^n)^(1/(2*n))*(a2+b2*x^n)^(1/(2*n)))] /;
FreeQ[{a1,b1,a2,b2},x] && EqQ[a2*b1+a1*b2,0] && IGtQ[2*n,0] && LtQ[-1,p,0] && NeQ[p,-1/2] &&
  IntegerQ[m] && LtQ[Denominator[p+(m+1)/(2*n)],Denominator[p]]
```

2. $\int (c\,x)^m \left(a+b\,x^n\right)^p dx$ when $n \in \mathbb{Z}^-$

   1. $\int (c\,x)^m \left(a+b\,x^n\right)^p dx$ when $n \in \mathbb{Z}^- \land m \in \mathbb{Q}$

      1: $\int x^m \left(a+b\,x^n\right)^p dx$ when $n \in \mathbb{Z}^- \land m \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z} \land m \in \mathbb{Z}$, then $x^m\,F[x^n] = -\text{Subst}\left[\frac{F[x^{-n}]}{x^{m+2}},\ x,\ \frac{1}{x}\right]\partial_x \frac{1}{x}$

Rule 1.1.3.2.7.2.1.1: If $n \in \mathbb{Z}^- \land m \in \mathbb{Z}$, then

$$\int x^m \left(a+b\,x^n\right)^p dx \longrightarrow -\text{Subst}\left[\int \frac{\left(a+b\,x^{-n}\right)^p}{x^{m+2}}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  -Subst[Int[(a+b*x^(-n))^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,p},x] && ILtQ[n,0] && IntegerQ[m]
```

```
Int[x_^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  -Subst[Int[(a1+b1*x^(-n))^p*(a2+b2*x^(-n))^p/x^(m+2),x],x,1/x] /;
FreeQ[{a1,b1,a2,b2,p},x] && EqQ[a2*b1+a1*b2,0] && ILtQ[2*n,0] && IntegerQ[m]
```

2: $\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx$ when $n \in \mathbb{Z}^- \land m \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $n \in \mathbb{Z} \land k > 1$, then $(c\,x)^m\,F[x^n] = -\frac{k}{c}\,\text{Subst}\left[\frac{F\left[c^{-n}\,x^{-k\,n}\right]}{x^{k\,(m+1)+1}},\,x,\,\frac{1}{(c\,x)^{1/k}}\right]\partial_x\frac{1}{(c\,x)^{1/k}}$

Rule 1.1.3.2.7.2.1.2: If $n \in \mathbb{Z}^- \land m \in \mathbb{F}$, let $k = \text{Denominator}[m]$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx \;\rightarrow\; -\frac{k}{c}\,\text{Subst}\left[\int \frac{\left(a+b\,c^{-n}\,x^{-k\,n}\right)^p}{x^{k\,(m+1)+1}}\,dx,\,x,\,\frac{1}{(c\,x)^{1/k}}\right]$$

Program code:

```
Int[(c_.*x_)^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  -k/c*Subst[Int[(a+b*c^(-n)*x^(-k*n))^p/x^(k*(m+1)+1),x],x,1/(c*x)^(1/k)]] /;
FreeQ[{a,b,c,p},x] && ILtQ[n,0] && FractionQ[m]
```

```
Int[(c_.*x_)^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  With[{k=Denominator[m]},
  -k/c*Subst[Int[(a1+b1*c^(-n)*x^(-k*n))^p*(a2+b2*c^(-n)*x^(-k*n))^p/x^(k*(m+1)+1),x],x,1/(c*x)^(1/k)]] /;
FreeQ[{a1,b1,a2,b2,c,p},x] && EqQ[a2*b1+a1*b2,0] && ILtQ[2*n,0] && FractionQ[m]
```

2: $\int (c\,x)^m \left(a + b\,x^n\right)^p \,dx$ when $n \in \mathbb{Z}^- \land m \notin \mathbb{Q}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \left((c\,x)^m \left(x^{-1}\right)^m\right) == 0$

Basis: $F[x] == -\text{Subst}\left[\frac{F\left[x^{-1}\right]}{x^2}, x, \frac{1}{x}\right] \partial_x \frac{1}{x}$

Rule 1.1.3.2.7.2.2: If $n \in \mathbb{Z}^- \land m \notin \mathbb{Q}$, then

$$\int (c\,x)^m \left(a + b\,x^n\right)^p \,dx \to (c\,x)^m \left(\frac{1}{x}\right)^m \int \frac{\left(a + b\,x^n\right)^p}{\left(\frac{1}{x}\right)^m} \,dx \to -\frac{1}{c} (c\,x)^{m+1} \left(\frac{1}{x}\right)^{m+1} \text{Subst}\left[\int \frac{\left(a + b\,x^{-n}\right)^p}{x^{m+2}} \,dx, x, \frac{1}{x}\right]$$

Program code:

```
Int[(c_.*x_)^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  -1/c*(c*x)^(m+1)*(1/x)^(m+1)*Subst[Int[(a+b*x^(-n))^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,b,c,m,p},x] && ILtQ[n,0] && Not[RationalQ[m]]
```

```
Int[(c_.*x_)^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  -1/c*(c*x)^(m+1)*(1/x)^(m+1)*Subst[Int[(a1+b1*x^(-n))^p*(a2+b2*x^(-n))^p/x^(m+2),x],x,1/x] /;
FreeQ[{a1,b1,a2,b2,c,m,p},x] && EqQ[a2*b1+a1*b2,0] && ILtQ[2*n,0] && Not[RationalQ[m]]
```

8. $\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx$ when $n \in \mathbb{F}$

1: $\int x^m \left(a+b\,x^n\right)^p\,dx$ when $n \in \mathbb{F}$

Derivation: Integration by substitution

Basis: If $k \in \mathbb{Z}^+$, then $x^m\,F[x^n] = k\,\text{Subst}\left[x^{k\,(m+1)-1}\,F\left[x^{k\,n}\right],\,x,\,x^{1/k}\right]\,\partial_x x^{1/k}$

Rule 1.1.3.2.8.1: If $n \in \mathbb{F}$, let $k = \text{Denominator}[n]$, then

$$\int x^m \left(a+b\,x^n\right)^p\,dx \;\longrightarrow\; k\,\text{Subst}\left[\int x^{k\,(m+1)-1}\left(a+b\,x^{k\,n}\right)^p\,dx,\,x,\,x^{1/k}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{k=Denominator[n]},
  k*Subst[Int[x^(k*(m+1)-1)*(a+b*x^(k*n))^p,x],x,x^(1/k)]] /;
FreeQ[{a,b,m,p},x] && FractionQ[n]
```

```
Int[x_^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  With[{k=Denominator[2*n]},
  k*Subst[Int[x^(k*(m+1)-1)*(a1+b1*x^(k*n))^p*(a2+b2*x^(k*n))^p,x],x,x^(1/k)]] /;
FreeQ[{a1,b1,a2,b2,m,p},x] && EqQ[a2*b1+a1*b2,0] && FractionQ[2*n]
```

2: $\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx$ when $n \in \mathbb{F}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m} == 0$

Rule 1.1.3.2.8.2: If $n \in \mathbb{F}$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx \longrightarrow \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a+b\,x^n\right)^p\,dx$$

Program code:

```
Int[(c_*x_)^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,m,p},x] && FractionQ[n]
```

```
Int[(c_*x_)^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a1+b1*x^n)^p*(a2+b2*x^n)^p,x] /;
FreeQ[{a1,b1,a2,b2,c,m,p},x] && EqQ[a2*b1+a1*b2,0] && FractionQ[2*n]
```

9. $\int (c\,x)^m \left(a+b\,x^n\right)^p \, dx$ when $\frac{n}{m+1} \in \mathbb{Z}$

1: $\int x^m \left(a+b\,x^n\right)^p \, dx$ when $\frac{n}{m+1} \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{n}{m+1} \in \mathbb{Z}$, then $x^m\,F[x^n] = \frac{1}{m+1}\,\text{Subst}\left[F\left[x^{\frac{n}{m+1}}\right], x, x^{m+1}\right] \partial_x x^{m+1}$

Rule 1.1.3.2.9.1: If $\frac{n}{m+1} \in \mathbb{Z}$, then

$$\int x^m \left(a+b\,x^n\right)^p dx \longrightarrow \frac{1}{m+1}\,\text{Subst}\left[\int \left(a+b\,x^{\frac{n}{m+1}}\right)^p dx,\ x,\ x^{m+1}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  1/(m+1)*Subst[Int[(a+b*x^Simplify[n/(m+1)])^p,x],x,x^(m+1)] /;
FreeQ[{a,b,m,n,p},x] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

```
Int[x_^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  1/(m+1)*Subst[Int[(a1+b1*x^Simplify[n/(m+1)])^p*(a2+b2*x^Simplify[n/(m+1)])^p,x],x,x^(m+1)] /;
FreeQ[{a1,b1,a2,b2,m,n,p},x] && EqQ[a2*b1+a1*b2,0] && IntegerQ[Simplify[2*n/(m+1)]] && Not[IntegerQ[2*n]]
```

2: $\int (c\,x)^m \left(a+b\,x^n\right)^p dx$ when $\frac{n}{m+1} \in \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m} == 0$

Rule 1.1.3.2.9.2: If $\frac{n}{m+1} \in \mathbb{Z}$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p dx \longrightarrow \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int x^m \left(a+b\,x^n\right)^p dx$$

Program code:

```
Int[(c_*x_)^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,m,n,p},x] && IntegerQ[Simplify[n/(m+1)]] && Not[IntegerQ[n]]
```

```
Int[(c_*x_)^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a1+b1*x^n)^p*(a2+b2*x^n)^p,x] /;
FreeQ[{a1,b1,a2,b2,c,m,n,p},x] && EqQ[a2*b1+a1*b2,0] && IntegerQ[Simplify[2*n/(m+1)]] && Not[IntegerQ[2*n]]
```

10. $\int (c\,x)^m\,(a+b\,x^n)^p\,dx$ when $\frac{m+1}{n}+p\in\mathbb{Z}$

1. $\int (c\,x)^m\,(a+b\,x^n)^p\,dx$ when $\frac{m+1}{n}+p\in\mathbb{Z}\,\land\,p>0$

1. $\int (c\,x)^m\,(a+b\,x^n)^p\,dx$ when $\frac{m+1}{n}+p==0\,\land\,p>0$

1: $\int x^m\,(a+b\,x^n)^p\,dx$ when $\frac{m+1}{n}+p==0\,\land\,p>0$

Reference: G&R 2.110.3

Derivation: Binomial recurrence 1a

Derivation: Integration by parts

Rule 1.1.3.2.10.1.1.1: If $\frac{m+1}{n}+p==0\,\land\,p>0$, then

$$\int x^m\,(a+b\,x^n)^p\,dx\,\longrightarrow\,\frac{x^{m+1}\,(a+b\,x^n)^p}{m+1}-\frac{b\,n\,p}{m+1}\int x^{m+n}\,(a+b\,x^n)^{p-1}\,dx$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  x^(m+1)*(a+b*x^n)^p/(m+1) -
  b*n*p/(m+1)*Int[x^(m+n)*(a+b*x^n)^(p-1),x] /;
FreeQ[{a,b,m,n},x] && EqQ[(m+1)/n+p,0] && GtQ[p,0]
```

```
Int[x_^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  x^(m+1)*(a1+b1*x^n)^p*(a2+b2*x^n)^p/(m+1) -
  2*b1*b2*n*p/(m+1)*Int[x^(m+2*n)*(a1+b1*x^n)^(p-1)*(a2+b2*x^n)^(p-1),x] /;
FreeQ[{a1,b1,a2,b2,m,n},x] && EqQ[a2*b1+a1*b2,0] && EqQ[(m+1)/(2*n)+p,0] && GtQ[p,0]
```

2: $\int (c\,x)^m\,(a+b\,x^n)^p\,dx$ when $\frac{m+1}{n}+p==0\land p>0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m}==0$

Rule 1.1.3.2.10.1.1.2: If $\frac{m+1}{n}+p==0\ \land\ p>0$, then

$$\int (c\,x)^m\,(a+b\,x^n)^p\,dx\ \longrightarrow\ \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[(c_*x_)^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,m,n},x] && EqQ[(m+1)/n+p,0] && GtQ[p,0]
```

```
Int[(c_*x_)^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a1+b1*x^n)^p*(a2+b2*x^n)^p,x] /;
FreeQ[{a1,b1,a2,b2,c,m,n},x] && EqQ[a2*b1+a1*b2,0] && EqQ[(m+1)/(2*n)+p,0] && GtQ[p,0]
```

2: $\int (c\,x)^m\,(a+b\,x^n)^p\,dx$ when $\frac{m+1}{n}+p\in\mathbb{Z}\land p>0\land m+n\,p+1\neq0$

Reference: G&R 2.110.1, CRC 88b

Derivation: Binomial recurrence 1b

Derivation: Inverted integration by parts

Rule 1.1.3.2.10.1.2: If $\frac{m+1}{n}+p\in\mathbb{Z}\ \land\ p>0\ \land\ m+n\,p+1\neq0$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p dx \longrightarrow \frac{(c\,x)^{m+1} \left(a+b\,x^n\right)^p}{c\,(m+n\,p+1)} + \frac{a\,n\,p}{m+n\,p+1} \int (c\,x)^m \left(a+b\,x^n\right)^{p-1} dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a+b*x^n)^p/(c*(m+n*p+1)) +
  a*n*p/(m+n*p+1)*Int[(c*x)^m*(a+b*x^n)^(p-1),x] /;
FreeQ[{a,b,c,m,n},x] && IntegerQ[p+Simplify[(m+1)/n]] && GtQ[p,0] && NeQ[m+n*p+1,0]
```

```
Int[(c_.*x_)^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a1+b1*x^n)^p*(a2+b2*x^n)^p/(c*(m+2*n*p+1)) +
  2*a1*a2*n*p/(m+2*n*p+1)*Int[(c*x)^m*(a1+b1*x^n)^(p-1)*(a2+b2*x^n)^(p-1),x] /;
FreeQ[{a1,b1,a2,b2,c,m,n},x] && EqQ[a2*b1+a1*b2,0] && IntegerQ[p+Simplify[(m+1)/(2*n)]] && GtQ[p,0] && NeQ[m+2*n*p+1,0]
```

2. $\int (c\,x)^m \left(a+b\,x^n\right)^p dx$ when $\frac{m+1}{n}+p\in\mathbb{Z}\land p<0$

   1. $\int (c\,x)^m \left(a+b\,x^n\right)^p dx$ when $\frac{m+1}{n}+p\in\mathbb{Z}\land -1<p<0$

      1: $\int x^m \left(a+b\,x^n\right)^p dx$ when $\frac{m+1}{n}+p\in\mathbb{Z}\land -1<p<0$

Derivation: Integration by substitution

Basis: If $\frac{m+1}{n}+p\in\mathbb{Z}$, let $k=\text{Denominator}[p]$, then

$$x^m\,(a+b\,x^n)^p = \frac{k\,a^{p+\frac{m+1}{n}}}{n}\,\text{Subst}\left[\frac{x^{\frac{k\,(m+1)}{n}-1}}{\left(1-b\,x^k\right)^{p+\frac{m+1}{n}+1}},\ x,\ \frac{x^{n/k}}{(a+b\,x^n)^{1/k}}\right]\partial_x\frac{x^{n/k}}{(a+b\,x^n)^{1/k}}$$

Basis: If $a2\,b1+a1\,b2==0\land\frac{m+1}{2\,n}+p\in\mathbb{Z}$, let $k=\text{Denominator}[p]$, then

$$x^m\,(a1+b1\,x^n)^p\,(a2+b2\,x^n)^p ==$$
$$\frac{k\,(a1\,a2)^{p+\frac{m+1}{2\,n}}}{2\,n}\,\text{Subst}\left[\frac{x^{\frac{k\,(m+1)}{2\,n}-1}}{\left(1-b1\,b2\,x^k\right)^{p+\frac{m+1}{2\,n}+1}},\ x,\ \frac{x^{2\,n/k}}{(a1+b1\,x^n)^{1/k}\,(a2+b2\,x^n)^{1/k}}\right]\partial_x\frac{x^{2\,n/k}}{(a1+b1\,x^n)^{1/k}\,(a2+b2\,x^n)^{1/k}}$$

Note: The exponents in the resulting integrand are integers.

Rule 1.1.3.2.10.2.1.1: If $\frac{m+1}{n}+p\in\mathbb{Z}\land -1<p<0$, let $k=\text{Denominator}[p]$, then

$$\int x^m\left(a+b\,x^n\right)^p dx \longrightarrow \frac{k\,a^{p+\frac{m+1}{n}}}{n}\,\text{Subst}\left[\int\frac{x^{\frac{k\,(m+1)}{n}-1}}{\left(1-b\,x^k\right)^{p+\frac{m+1}{n}+1}}dx,\ x,\ \frac{x^{n/k}}{\left(a+b\,x^n\right)^{1/k}}\right]$$

$$\int x^m\left(a1+b1\,x^n\right)^p\left(a2+b2\,x^n\right)^p dx \longrightarrow \frac{k\,(a1\,a2)^{p+\frac{m+1}{2\,n}}}{2\,n}\,\text{Subst}\left[\int\frac{x^{\frac{k\,(m+1)}{2\,n}-1}}{\left(1-b1\,b2\,x^k\right)^{p+\frac{m+1}{2\,n}+1}}dx,\ x,\ \frac{x^{2\,n/k}}{\left(a1+b1\,x^n\right)^{1/k}\left(a2+b2\,x^n\right)^{1/k}}\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  With[{k=Denominator[p]},
  k*a^(p+Simplify[(m+1)/n])/n*
    Subst[Int[x^(k*Simplify[(m+1)/n]-1)/(1-b*x^k)^(p+Simplify[(m+1)/n]+1),x],x,x^(n/k)/(a+b*x^n)^(1/k)]] /;
FreeQ[{a,b,m,n},x] && IntegerQ[p+Simplify[(m+1)/n]] && LtQ[-1,p,0]
```

```
Int[x_^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  With[{k=Denominator[p]},
  k*(a1*a2)^(p+Simplify[(m+1)/(2*n)])/(2*n)*
    Subst[Int[x^(k*Simplify[(m+1)/(2*n)]-1)/(1-b1*b2*x^k)^(p+Simplify[(m+1)/(2*n)]+1),x],x,x^(2*n/k)/((a1+b1*x^n)^(1/k)*(a2+b2*x^n)^
FreeQ[{a1,b1,a2,b2,m,n},x] && EqQ[a2*b1+a1*b2,0] && IntegerQ[p+Simplify[(m+1)/(2*n)]] && LtQ[-1,p,0]
```

2: $\int (c\,x)^m\,(a+b\,x^n)^p\,dx$ when $\frac{m+1}{n}+p\in\mathbb{Z}\ \land\ -1<p<0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c\,x)^m}{x^m} == 0$

Rule 1.1.3.2.10.2.1.2: If $\frac{m+1}{n}+p\in\mathbb{Z}\ \land\ -1<p<0$, then

$$\int (c\,x)^m\,(a+b\,x^n)^p\,dx \longrightarrow \frac{c^{\text{IntPart}[m]}\,(c\,x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}}\int x^m\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[(c_*x_)^m_*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,m,n},x] && IntegerQ[p+Simplify[(m+1)/n]] && LtQ[-1,p,0]
```

```
Int[(c_*x_)^m_*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m*(a1+b1*x^n)^p*(a2+b2*x^n)^p,x] /;
FreeQ[{a1,b1,a2,b2,c,m,n},x] && EqQ[a2*b1+a1*b2,0] && IntegerQ[p+Simplify[(m+1)/(2*n)]] && LtQ[-1,p,0]
```

2: $\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx$ when $\frac{m+1}{n}+p\in\mathbb{Z}\,\land\,p<-1$

Reference: G&R 2.110.2, CRC 88d

Derivation: Binomial recurrence 2b

Derivation: Integration by parts

Basis: $x^m\,(a+b\,x^n)^p == x^{m+n\,p+n+1}\,\frac{(a+b\,x^n)^p}{x^{n\,(p+1)+1}}$

Basis: $\int \frac{(a+b\,x^n)^p}{x^{n\,(p+1)+1}}\,dx == -\frac{(a+b\,x^n)^{p+1}}{x^{n\,(p+1)}\,a\,n\,(p+1)}$

Rule 1.1.3.2.10.2.2: If $\frac{m+1}{n}+p\in\mathbb{Z}\,\land\,p<-1$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx \longrightarrow -\frac{(c\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}}{a\,c\,n\,(p+1)}+\frac{m+n\,(p+1)+1}{a\,n\,(p+1)}\int (c\,x)^m\left(a+b\,x^n\right)^{p+1}dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  -(c*x)^(m+1)*(a+b*x^n)^(p+1)/(a*c*n*(p+1)) +
  (m+n*(p+1)+1)/(a*n*(p+1))*Int[(c*x)^m*(a+b*x^n)^(p+1),x] /;
FreeQ[{a,b,c,m,n},x] && IntegerQ[p+Simplify[(m+1)/n]] && LtQ[p,-1]
```

```
Int[(c_.*x_)^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  -(c*x)^(m+1)*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(2*a1*a2*c*n*(p+1)) +
  (m+2*n*(p+1)+1)/(2*a1*a2*n*(p+1))*Int[(c*x)^m*(a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1),x] /;
FreeQ[{a1,b1,a2,b2,c,m,n},x] && EqQ[a2*b1+a1*b2,0] && IntegerQ[p+Simplify[(m+1)/(2*n)]] && LtQ[p,-1]
```

11. $\int \frac{(c\,x)^m}{a+b\,x^n}\,dx$ when $\frac{m+1}{n} \in \mathbb{F}$

1. $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{m+1}{n} \in \mathbb{F}$

1: $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{m+1}{n} \in \mathbb{F} \land m-n \lessdot m$

Reference: CRC 86

Derivation: Binomial recurrence 3a with $p = -1$

Rule 1.1.3.2.11.1.1: If $\frac{m+1}{n} \in \mathbb{F} \land m-n \lessdot m$, then

$$\int \frac{x^m}{a+b\,x^n}\,dx \longrightarrow \frac{x^{m-n+1}}{b\,(m-n+1)} - \frac{a}{b}\int \frac{x^{m-n}}{a+b\,x^n}\,dx$$

Program code:

```
Int[x_^m_./(a_+b_.*x_^n_),x_Symbol] :=
  With[{mn=Simplify[m-n]},
  x^(mn+1)/(b*(mn+1)) -
  a/b*Int[x^mn/(a+b*x^n),x]] /;
FreeQ[{a,b,m,n},x] && FractionQ[Simplify[(m+1)/n]] && SumSimplerQ[m,-n]
```

2: $\int \frac{x^m}{a+b\,x^n}\,dx$ when $\frac{m+1}{n} \in \mathbb{F} \land m+n \lessdot m$

Reference: CRC 87

Derivation: Binomial recurrence 3b with $p = -1$

Rule 1.1.3.2.11.1.2: If $\frac{m+1}{n} \in \mathbb{F} \land m+n \lessdot m$, then

$$\int \frac{x^m}{a+b x^n} \, dx \longrightarrow \frac{x^{m+1}}{a (m+1)} - \frac{b}{a} \int \frac{x^{m+n}}{a+b x^n} \, dx$$

Program code:

```
Int[x_^m_/(a_+b_.*x_^n_),x_Symbol] :=
  x^(m+1)/(a*(m+1)) -
  b/a*Int[x^Simplify[m+n]/(a+b*x^n),x] /;
FreeQ[{a,b,m,n},x] && FractionQ[Simplify[(m+1)/n]] && SumSimplerQ[m,n]
```

2: $\int \frac{(c x)^m}{a+b x^n} \, dx$ when $\frac{m+1}{n} \in \mathbb{F}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c x)^m}{x^m} == 0$

Rule 1.1.3.2.11.2: If $\frac{m+1}{n} \in \mathbb{F}$, then

$$\int \frac{(c x)^m}{a+b x^n} \, dx \longrightarrow \frac{c^{\text{IntPart}[m]} (c x)^{\text{FracPart}[m]}}{x^{\text{FracPart}[m]}} \int \frac{x^m}{a+b x^n} \, dx$$

Program code:

```
Int[(c_*x_)^m_/(a_+b_.*x_^n_),x_Symbol] :=
  c^IntPart[m]*(c*x)^FracPart[m]/x^FracPart[m]*Int[x^m/(a+b*x^n),x] /;
FreeQ[{a,b,c,m,n},x] && FractionQ[Simplify[(m+1)/n]] && (SumSimplerQ[m,n] || SumSimplerQ[m,-n])
```

12. $\int (c\,x)^m \left(a+b\,x^n\right)^p \, dx$ when $p \notin \mathbb{Z}^+$

1: $\int (c\,x)^m \left(a+b\,x^n\right)^p \, dx$ when $p \notin \mathbb{Z}^+ \land (p \in \mathbb{Z}^- \lor a > 0)$

Note: If $t == r+1 \land r \in \mathbb{Z}$, then **Hypergeometric2F1[r, s, t, z] == Hypergeometric2F1[s, r, t, z]** are elementary or undefined.

Rule 1.1.3.2.12.1: If $p \notin \mathbb{Z}^+ \land (p \in \mathbb{Z}^- \lor a > 0)$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p \, dx \to \frac{a^p\,(c\,x)^{m+1}}{c\,(m+1)}\,\text{Hypergeometric2F1}\left[-p,\,\frac{m+1}{n},\,\frac{m+1}{n}+1,\,-\frac{b\,x^n}{a}\right]$$

Program code:

```
Int[(c_.*x_)^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  a^p*(c*x)^(m+1)/(c*(m+1))*Hypergeometric2F1[-p,(m+1)/n,(m+1)/n+1,-b*x^n/a] /;
FreeQ[{a,b,c,m,n,p},x] && Not[IGtQ[p,0]] && (ILtQ[p,0] || GtQ[a,0])
```

x: $\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx$ when $p\notin\mathbb{Z}^+ \land \neg\,(p\in\mathbb{Z}^- \lor a>0)$

Note: If $r==1 \land (s\in\mathbb{Z} \lor t\in\mathbb{Z})$, then **Hypergeometric2F1[r, s, t, z] == Hypergeometric2F1[s, r, t, z]** are undefined or can be expressed in elementary form.

Note: *Mathematica* has a hard time simplifying the derivative of the following antiderivative to the integrand, so the following, more complicated, but easily differentiated, rule is used instead.

Rule 1.1.3.2.12.x: If $p\notin\mathbb{Z}^+ \land \neg\,(p\in\mathbb{Z}^- \lor a>0)$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx \;\rightarrow\; \frac{(c\,x)^{m+1}\left(a+b\,x^n\right)^{p+1}}{a\,c\,(m+1)}\,\text{Hypergeometric2F1}\left[1,\,\frac{m+1}{n}+p+1,\,\frac{m+1}{n}+1,\,-\frac{b\,x^n}{a}\right]$$

Program code:

```
(* Int[(c_.*x_)^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  (c*x)^(m+1)*(a+b*x^n)^(p+1)/(a*c*(m+1))*Hypergeometric2F1[1,(m+1)/n+p+1,(m+1)/n+1,-b*x^n/a] /;
FreeQ[{a,b,c,m,n,p},x] && Not[IGtQ[p,0]] && Not[ILtQ[p,0] || GtQ[a,0]] *)
```

2: $\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx$ when $p \notin \mathbb{Z}^+ \land \neg\,(p \in \mathbb{Z}^- \lor a>0)$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a+b\,x^n)^p}{\left(1+\frac{b\,x^n}{a}\right)^p} == 0$

Rule 1.1.3.2.12.2: If $p \notin \mathbb{Z}^+ \land \neg\,(p \in \mathbb{Z}^- \lor a>0)$, then

$$\int (c\,x)^m \left(a+b\,x^n\right)^p\,dx \longrightarrow \frac{a^{\text{IntPart}[p]}\left(a+b\,x^n\right)^{\text{FracPart}[p]}}{\left(1+\frac{b\,x^n}{a}\right)^{\text{FracPart}[p]}} \int (c\,x)^m \left(1+\frac{b\,x^n}{a}\right)^p\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a_+b_.*x_^n_)^p_,x_Symbol] :=
  a^IntPart[p]*(a+b*x^n)^FracPart[p]/(1+b*x^n/a)^FracPart[p]*Int[(c*x)^m*(1+b*x^n/a)^p,x] /;
FreeQ[{a,b,c,m,n,p},x] && Not[IGtQ[p,0]] && Not[ILtQ[p,0] || GtQ[a,0]]
```

D: $\int (c\,x)^m \left(a_1+b_1\,x^n\right)^p \left(a_2+b_2\,x^n\right)^p\,dx$ when $a_2\,b_1+a_1\,b_2==0\land p\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $a_2\,b_1+a_1\,b_2==0$, then $\partial_x \frac{\left(a_1+b_1\,x^n\right)^p\left(a_2+b_2\,x^n\right)^p}{\left(a_1\,a_2+b_1\,b_2\,x^{2\,n}\right)^p}==0$

Rule: If $a_2\,b_1+a_1\,b_2==0\land p\notin\mathbb{Z}$, then

$$\int (c\,x)^m \left(a_1+b_1\,x^n\right)^p \left(a_2+b_2\,x^n\right)^p\,dx \to \frac{\left(a_1+b_1\,x^n\right)^{\text{FracPart}[p]}\left(a_2+b_2\,x\right)^{\text{FracPart}[p]}}{\left(a_1\,a_2+b_1\,b_2\,x^2\right)^{\text{FracPart}[p]}}\int (c\,x)^m\left(a_1\,a_2+b_1\,b_2\,x^{2\,n}\right)^p\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*(a1_+b1_.*x_^n_)^p_*(a2_+b2_.*x_^n_)^p_,x_Symbol] :=
  (a1+b1*x^n)^FracPart[p]*(a2+b2*x^n)^FracPart[p]/(a1*a2+b1*b2*x^(2*n))^FracPart[p]*Int[(c*x)^m*(a1*a2+b1*b2*x^(2*n))^p,x] /;
FreeQ[{a1,b1,a2,b2,c,m,n,p},x] && EqQ[a2*b1+a1*b2,0] && Not[IntegerQ[p]]
```

```
(* IntBinomialQ[a,b,c,n,m,p,x] returns True iff (c*x)^m*(a+b*x^n)^p is integrable wrt x in terms of non-hypergeometric functions. *)
IntBinomialQ[a_,b_,c_,n_,m_,p_,x_] :=
  IGtQ[p,0] || RationalQ[m] && IntegersQ[n,2*p] || IntegerQ[(m+1)/n+p] ||
  (EqQ[n,2] || EqQ[n,4]) && IntegersQ[2*m,4*p] ||
  EqQ[n,2] && IntegerQ[6*p] && (IntegerQ[m] || IntegerQ[m-p])
```

Rules for integrands of the form $(d\,x)^m\,(a+b\,(c\,x^q)^n)^p$

1: $\int (d\,x)^m\,(a+b\,(c\,x)^n)^p\,dx$

Derivation: Integration by substitution

Rule:

$$\int (d\,x)^m\,(a+b\,(c\,x)^n)^p\,dx \longrightarrow \frac{1}{c}\,\text{Subst}\!\left[\int \left(\frac{d\,x}{c}\right)^m (a+b\,x^n)^p\,dx,\ x,\ c\,x\right]$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*(c_*x_)^n_)^p_.,x_Symbol] :=
  1/c*Subst[Int[(d*x/c)^m*(a+b*x^n)^p,x],x,c*x] /;
FreeQ[{a,b,c,d,m,n,p},x]
```

2: $\int (d\,x)^m\,\left(a+b\,\left(c\,x^q\right)^n\right)^p\,dx$ when $n\,q\in\mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis: $\partial_x \frac{(d\,x)^{m+1}}{\left(\left(c\,x^q\right)^{1/q}\right)^{m+1}} == 0$

Basis: $\frac{F\left[\left(c\,x^q\right)^{1/q}\right]}{x} == \text{Subst}\left[\frac{F[x]}{x},\,x,\,\left(c\,x^q\right)^{1/q}\right]\,\partial_x \left(c\,x^q\right)^{1/q}$

Rule: If $n\,q\in\mathbb{Z}$, then

$$\int (d\,x)^m\,\left(a+b\,\left(c\,x^q\right)^n\right)^p\,dx \;\rightarrow\; \frac{(d\,x)^{m+1}}{d\,\left(\left(c\,x^q\right)^{1/q}\right)^{m+1}}\int \frac{\left(\left(c\,x^q\right)^{1/q}\right)^{m+1}\,\left(a+b\,\left(\left(c\,x^q\right)^{1/q}\right)^{n\,q}\right)^p}{x}\,dx$$

$$\rightarrow\; \frac{(d\,x)^{m+1}}{d\,\left(\left(c\,x^q\right)^{1/q}\right)^{m+1}}\,\text{Subst}\left[\int x^m\,\left(a+b\,x^{n\,q}\right)^p\,dx,\,x,\,\left(c\,x^q\right)^{1/q}\right]$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*(c_.*x_^q_)^n_)^p_.,x_Symbol] :=
  (d*x)^(m+1)/(d*((c*x^q)^(1/q))^(m+1))*Subst[Int[x^m*(a+b*x^(n*q))^p,x],x,(c*x^q)^(1/q)] /;
FreeQ[{a,b,c,d,m,n,p,q},x] && IntegerQ[n*q] && NeQ[x,(c*x^q)^(1/q)]
```

3: $\int (d\,x)^m \left(a+b\left(c\,x^q\right)^n\right)^p \,dx$ when $n \in \mathbb{F}$

Derivation: Integration by substitution

Rule: If $n \in \mathbb{F}$, then

$$\int (d\,x)^m \left(a+b\left(c\,x^q\right)^n\right)^p \,dx \to \text{Subst}\left[\int (d\,x)^m \left(a+b\,c^n\,x^{n\,q}\right)^p \,dx,\ x^{1/k},\ \frac{\left(c\,x^q\right)^{1/k}}{c^{1/k}\left(x^{1/k}\right)^{q-1}}\right]$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*(c_.*x_^q_)^n_)^p_.,x_Symbol] :=
  With[{k=Denominator[n]},
  Subst[Int[(d*x)^m*(a+b*c^n*x^(n*q))^p,x],x^(1/k),(c*x^q)^(1/k)/(c^(1/k)*(x^(1/k))^(q-1))]] /;
FreeQ[{a,b,c,d,m,p,q},x] && FractionQ[n]
```

4: $\int (d\,x)^m \left(a+b\left(c\,x^q\right)^n\right)^p \,dx$ when $n \notin \mathbb{R}$

Derivation: Integration by substitution

Basis: $F\left[\left(c\,x^q\right)^n\right] = \text{Subst}\left[F\left[c^n\,x^{n\,q}\right],\,x^{n\,q},\,\frac{\left(c\,x^q\right)^n}{c^n}\right]$

Rule: If $n \notin \mathbb{R}$, then

$$\int (d\,x)^m \left(a+b\left(c\,x^q\right)^n\right)^p \,dx \,\rightarrow\, \text{Subst}\left[\int (d\,x)^m \left(a+b\,c^n\,x^{n\,q}\right)^p \,dx,\,x^{n\,q},\,\frac{\left(c\,x^q\right)^n}{c^n}\right]$$

Program code:

```
Int[(d_.*x_)^m_.*(a_+b_.*(c_.*x_^q_)^n_)^p_.,x_Symbol] :=
  Subst[Int[(d*x)^m*(a+b*c^n*x^(n*q))^p,x],x^(n*q),(c*x^q)^n/c^n] /;
FreeQ[{a,b,c,d,m,n,p,q},x] && Not[RationalQ[n]]
```

S. $\int u^m \left(a+b\,v^n\right)^p\,dx$

1: $\int x^m \left(a+b\,v^n\right)^p\,dx$ when $v == c+d\,x \land m \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $m \in \mathbb{Z}$, then $x^m\,F[c+d\,x] == \frac{1}{d^{m+1}}\,\text{Subst}\left[(x-c)^m\,F[x],\,x,\,c+d\,x\right]\,\partial_x\,(c+d\,x)$

Rule 1.1.3.2.S.2: If $v == c+d\,x \land m \in \mathbb{Z}$, then

$$\int x^m \left(a+b\,v^n\right)^p\,dx \longrightarrow \frac{1}{d^{m+1}}\,\text{Subst}\left[\int (x-c)^m \left(a+b\,x^n\right)^p\,dx,\,x,\,v\right]$$

Program code:

```
Int[x_^m_.*(a_+b_.*v_^n_)^p_.,x_Symbol] :=
  With[{c=Coefficient[v,x,0],d=Coefficient[v,x,1]},
  1/d^(m+1)*Subst[Int[SimplifyIntegrand[(x-c)^m*(a+b*x^n)^p,x],x],x,v] /;
 NeQ[c,0]] /;
FreeQ[{a,b,n,p},x] && LinearQ[v,x] && IntegerQ[m]
```

2: $\int u^m \left(a+b\,v^n\right)^p \, dx$ when $v == c+d\,x \land u == e\,v$

Derivation: Integration by substitution and piecewise constant extraction

Basis: If $u == e\,v$, then $\partial_x \frac{u^m}{v^m} == 0$

Rule 1.1.3.2.S.3: If $v == c+d\,x \land u == e\,v$, then

$$\int u^m \left(a+b\,v^n\right)^p \, dx \rightarrow \frac{u^m}{d\,v^m}\,\text{Subst}\left[\int x^m \left(a+b\,x^n\right)^p \, dx,\ x,\ v\right]$$

Program code:

```
Int[u_^m_.*(a_+b_.*v_^n_)^p_.,x_Symbol] :=
  u^m/(Coefficient[v,x,1]*v^m)*Subst[Int[x^m*(a+b*x^n)^p,x],x,v] /;
FreeQ[{a,b,m,n,p},x] && LinearPairQ[u,v,x]
```