Rules for integrands of the form $(a+b x)^m (c+d x)^n$

0: $\int (a+b x)^m (c+d x)\, dx$ when $a d-b c (m+2) == 0$

Derivation: Algebraic expansion

Basis: If $a d-b c (m+2) == 0$, then $c+d x == \frac{d (a+b (m+2) x)}{b (m+2)}$

Rule 1.1.1.2.0: If $a d-b c (m+2) == 0$, then

$$\int (a+b x)^m (c+d x)\, dx \longrightarrow \frac{d}{b (m+2)} \int (a+b x)^m (a+b (m+2) x)\, dx \longrightarrow \frac{d x (a+b x)^{m+1}}{b (m+2)}$$

```
Int[(a_+b_.*x_)^m_.*(c_+d_.*x_),x_Symbol] :=
  d*x*(a+b*x)^(m+1)/(b*(m+2)) /;
FreeQ[{a,b,c,d,m},x] && EqQ[a*d-b*c*(m+2),0]
```

1. $\int (a+b\,x)^m\,(c+d\,x)^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,m+n+2==0$

   1. $\int \frac{1}{(a+b\,x)\,(c+d\,x)}\,dx$ when $b\,c-a\,d\neq 0$

      1: $\int \frac{1}{(a+b\,x)\,(c+d\,x)}\,dx$ when $b\,c+a\,d==0$

Derivation: Algebraic simplification

Basis: If $b\,c+a\,d==0$, then $(a+b\,x)\,(c+d\,x)==a\,c+b\,d\,x^2$

Rule 1.1.1.2.1.1.1: If $b\,c+a\,d==0$, then

$$\int \frac{1}{(a+b\,x)\,(c+d\,x)}\,dx\,\longrightarrow\,\int \frac{1}{a\,c+b\,d\,x^2}\,dx$$

Program code:

```
Int[1/((a_+b_.*x_)*(c_+d_.*x_)),x_Symbol] :=
  Int[1/(a*c+b*d*x^2),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0]
```

2: $\int \frac{1}{(a+b\,x)\,(c+d\,x)}\,dx$ when $b\,c-a\,d\neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{(a+b\,x)\,(c+d\,x)} == \frac{b}{(b\,c-a\,d)\,(a+b\,x)} - \frac{d}{(b\,c-a\,d)\,(c+d\,x)}$

Rule 1.1.1.2.1.1.2: If $b\,c-a\,d\neq 0$, then

$$\int \frac{1}{(a+b\,x)\,(c+d\,x)}\,dx \longrightarrow \frac{b}{b\,c-a\,d}\int \frac{1}{a+b\,x}\,dx - \frac{d}{b\,c-a\,d}\int \frac{1}{c+d\,x}\,dx$$

Program code:

```
Int[1/((a_.+b_.*x_)*(c_.+d_.*x_)),x_Symbol] :=
  b/(b*c-a*d)*Int[1/(a+b*x),x] - d/(b*c-a*d)*Int[1/(c+d*x),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0]
```

2: $\int (a+b\,x)^m\,(c+d\,x)^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m+n+2==0\,\wedge\,m\neq -1$

Reference: G&R 2.155, CRC 59a with $m+n+2==0$

Reference: G&R 2.110.2 or 2.110.6 with $k=1$ and $m+n+2==0$

Derivation: Linear recurrence 3 with $m+n+2==0$

Rule 1.1.1.2.1.2: If $b\,c-a\,d\neq 0\,\wedge\,m+n+2==0\,\wedge\,m\neq -1$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx\,\longrightarrow\,\frac{(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}}{(b\,c-a\,d)\,(m+1)}$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^(n+1)/((b*c-a*d)*(m+1)) /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[b*c-a*d,0] && EqQ[m+n+2,0] && NeQ[m,-1]
```

2. $\int (a+b x)^m (c+d x)^n \, dx$ when $b c+a d == 0 \land n == m$

1: $\int (a+b x)^m (c+d x)^m \, dx$ when $b c+a d == 0 \land m+\frac{1}{2} \in \mathbb{Z}^+$

Derivation: Inverted integration by parts

Rule 1.1.1.2.2.1: If $b c+a d == 0 \land m+\frac{1}{2} \in \mathbb{Z}^+$, then

$$\int (a+b x)^m (c+d x)^m \, dx \longrightarrow \frac{x (a+b x)^m (c+d x)^m}{2 m+1}+\frac{2 a c m}{2 m+1} \int (a+b x)^{m-1} (c+d x)^{m-1} \, dx$$

Program code:

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^m_,x_Symbol] :=
  x*(a+b*x)^m*(c+d*x)^m/(2*m+1) + 2*a*c*m/(2*m+1)*Int[(a+b*x)^(m-1)*(c+d*x)^(m-1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && IGtQ[m+1/2,0]
```

2. $\int (a+b x)^m (c+d x)^m \, dx$ when $b c+a d == 0 \land m+\frac{1}{2} \in \mathbb{Z}^-$

1: $\int \frac{1}{(a+b x)^{3/2} (c+d x)^{3/2}} \, dx$ when $b c+a d == 0$

Rule 1.1.1.2.2.2.1: If $b c+a d == 0$, then

$$\int \frac{1}{(a+b x)^{3/2} (c+d x)^{3/2}} \, dx \longrightarrow \frac{x}{a c \sqrt{a+b x} \sqrt{c+d x}}$$

Program code:

```
Int[1/((a_+b_.*x_)^(3/2)*(c_+d_.*x_)^(3/2)),x_Symbol] :=
  x/(a*c*Sqrt[a+b*x]*Sqrt[c+d*x]) /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0]
```

2: $\int (a+b\,x)^m\,(c+d\,x)^m\,dx$ when $b\,c+a\,d=0\,\land\,m+\frac{3}{2}\in\mathbb{Z}^-$

Derivation: Integration by parts

Basis: $(a+b\,x)^m\,(c+d\,x)^m = x^{2\,(m+1)+1}\,\frac{(a+b\,x)^m\,(c+d\,x)^m}{x^{2\,(m+1)+1}}$

Basis: If $b\,c+a\,d=0$, then $\int\frac{(a+b\,x)^m\,(c+d\,x)^m}{x^{2\,(m+1)+1}}\,dx = -\frac{(a+b\,x)^{m+1}\,(c+d\,x)^{m+1}}{x^{2\,(m+1)}\,2\,a\,c\,(m+1)}$

Rule 1.1.1.2.2.2.2: If $b\,c+a\,d=0\,\land\,m+\frac{3}{2}\in\mathbb{Z}^-$, then

$$\int (a+b\,x)^m\,(c+d\,x)^m\,dx \longrightarrow -\frac{x\,(a+b\,x)^{m+1}\,(c+d\,x)^{m+1}}{2\,a\,c\,(m+1)}+\frac{2\,m+3}{2\,a\,c\,(m+1)}\int (a+b\,x)^{m+1}\,(c+d\,x)^{m+1}\,dx$$

Program code:

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^m_,x_Symbol] :=
  -x*(a+b*x)^(m+1)*(c+d*x)^(m+1)/(2*a*c*(m+1)) +
  (2*m+3)/(2*a*c*(m+1))*Int[(a+b*x)^(m+1)*(c+d*x)^(m+1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && ILtQ[m+3/2,0]
```

3: $\int (a+b\,x)^m\,(c+d\,x)^m\,dx$ when $b\,c+a\,d==0\,\land\,(m\in\mathbb{Z}\,\lor\,a>0\,\land\,c>0)$

Derivation: Algebraic simplification

Basis: If $b\,c+a\,d==0\,\land\,(m\in\mathbb{Z}\,\lor\,a>0\,\land\,c>0)$, then $(a+b\,x)^m\,(c+d\,x)^m==(a\,c+b\,d\,x^2)^m$

Rule 1.1.1.2.2.3: If $b\,c+a\,d==0\,\land\,(m\in\mathbb{Z}\,\lor\,a>0\,\land\,c>0)$, then

$$\int (a+b\,x)^m\,(c+d\,x)^m\,dx\;\longrightarrow\;\int \left(a\,c+b\,d\,x^2\right)^m\,dx$$

Program code:

```
Int[(a_+b_.*x_)^m_.*(c_+d_.*x_)^m_.,x_Symbol] :=
  Int[(a*c+b*d*x^2)^m,x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[b*c+a*d,0] && (IntegerQ[m] || GtQ[a,0] && GtQ[c,0])
```

4: $\int (a+b\,x)^m\,(c+d\,x)^m\,dx$ when $b\,c+a\,d==0\,\wedge\,2\,m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b\,c+a\,d==0$, then $\partial_x \frac{(a+b\,x)^m\,(c+d\,x)^m}{\left(a\,c+b\,d\,x^2\right)^m}==0$

Basis: If $b\,c+a\,d==0$, then $\frac{(a+b\,x)^m\,(c+d\,x)^m}{\left(a\,c+b\,d\,x^2\right)^m}==\frac{(a+b\,x)^{\text{FracPart}[m]}\,(c+d\,x)^{\text{FracPart}[m]}}{\left(a\,c+b\,d\,x^2\right)^{\text{FracPart}[m]}}$

Rule 1.1.1.2.2.4: If $b\,c+a\,d==0\,\wedge\,2\,m\notin\mathbb{Z}$, then

$$\int (a+b\,x)^m\,(c+d\,x)^m\,dx\;\longrightarrow\;\frac{(a+b\,x)^{\text{FracPart}[m]}\,(c+d\,x)^{\text{FracPart}[m]}}{\left(a\,c+b\,d\,x^2\right)^{\text{FracPart}[m]}}\int\left(a\,c+b\,d\,x^2\right)^m\,dx$$

Program code:

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^m_,x_Symbol] :=
  (a+b*x)^FracPart[m]*(c+d*x)^FracPart[m]/(a*c+b*d*x^2)^FracPart[m]*Int[(a*c+b*d*x^2)^m,x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[b*c+a*d,0] && Not[IntegerQ[2*m]]
```

3: $\int (a + b x)^m (c + d x)^n \, dx$ when $b c - a d \neq 0 \land m \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule 1.1.1.2.3: If $b c - a d \neq 0 \land m \in \mathbb{Z}$, then

$$\int (a + b x)^m (c + d x)^n \, dx \longrightarrow \int \text{ExpandIntegrand}\left[(a + b x)^m (c + d x)^n, x\right] dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*(c_.+d_.*x_)^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)^m*(c+d*x)^n,x],x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && IGtQ[m,0] &&
  (Not[IntegerQ[n]] || EqQ[c,0] && LeQ[7*m+4*n+4,0] || LtQ[9*m+5*(n+1),0] || GtQ[m+n+2,0])
```

```
Int[(a_+b_.*x_)^m_.*(c_.+d_.*x_)^n_.,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*x)^m*(c+d*x)^n,x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && ILtQ[m,0] && IntegerQ[n] && Not[IGtQ[n,0] && LtQ[m+n+2,0]]
```

4: $\int (a+b\,x)^m\,(c+d\,x)^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,m+n+2\in\mathbb{Z}^-\,\land\,m\neq -1$

Reference: G&R 2.155, CRC 59a

Reference: G&R 2.110.2 or 2.110.6 with $k=1$

Derivation: Linear recurrence 3

Derivation: Integration by parts

Basis: $(a+b\,x)^m\,(c+d\,x)^n == (c+d\,x)^{m+n+2}\,\frac{(a+b\,x)^m}{(c+d\,x)^{m+2}}$

Rule 1.1.1.2.4: If $b\,c-a\,d\neq 0\,\land\,m+n+2\in\mathbb{Z}^-\,\land\,m\neq -1$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx\,\to\,\frac{(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}}{(b\,c-a\,d)\,(m+1)}-\frac{d\,(m+n+2)}{(b\,c-a\,d)\,(m+1)}\int (a+b\,x)^{m+1}\,(c+d\,x)^n\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^(n+1)/((b*c-a*d)*(m+1)) -
  d*Simplify[m+n+2]/((b*c-a*d)*(m+1))*Int[(a+b*x)^Simplify[m+1]*(c+d*x)^n,x] /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[b*c-a*d,0] && ILtQ[Simplify[m+n+2],0] && NeQ[m,-1] &&
  Not[LtQ[m,-1] && LtQ[n,-1] && (EqQ[a,0] || NeQ[c,0] && LtQ[m-n,0] && IntegerQ[n])] &&
  (SumSimplerQ[m,1] || Not[SumSimplerQ[n,1]])
```

5. $\int (a+b x)^m (c+d x)^n \, dx$ when $b c-a d \neq 0 \wedge n>0$

1: $\int (a+b x)^m (c+d x)^n \, dx$ when $b c-a d \neq 0 \wedge n>0 \wedge m<-1$

Reference: G&R 2.110.3 or 2.110.4 with $k=1$

Derivation: Linear recurrence 1

Note: If $n \in \mathbb{Z}$ and $m \notin \mathbb{Z}$, there is no need to drive m toward 0 along with n.

Rule 1.1.1.2.5.1: If $b c-a d \neq 0 \wedge n>0 \wedge m<-1$, then

$$\int (a+b x)^m (c+d x)^n \, dx \longrightarrow \frac{(a+b x)^{m+1} (c+d x)^n}{b (m+1)}-\frac{d n}{b (m+1)} \int (a+b x)^{m+1} (c+d x)^{n-1} \, dx$$

Program code:

```
Int[1/((a_+b_.*x_)^(9/4)*(c_+d_.*x_)^(1/4)),x_Symbol] :=
  -4/(5*b*(a+b*x)^(5/4)*(c+d*x)^(1/4)) - d/(5*b)*Int[1/((a+b*x)^(5/4)*(c+d*x)^(5/4)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && NegQ[a^2*b^2]
```

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n/(b*(m+1)) -
  d*n/(b*(m+1))*Int[(a+b*x)^(m+1)*(c+d*x)^(n-1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && GtQ[n,0] && LtQ[m,-1] && Not[IntegerQ[n] && Not[IntegerQ[m]]] &&
  Not[ILeQ[m+n+2,0] && (FractionQ[m] || GeQ[2*n+m+1,0])] && IntLinearQ[a,b,c,d,m,n,x]
```

2: $\int (a+b\,x)^m\,(c+d\,x)^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,n>0\,\land\,m+n+1\neq 0$

Reference: G&R 2.151, CRC 59b

Reference: G&R 2.110.1 or 2.110.5 with $k = 1$

Derivation: Linear recurrence 2

Derivation: Inverted integration by parts

Rule 1.1.1.2.5.2: If $b\,c-a\,d\neq 0\,\land\,n>0\,\land\,m+n+1\neq 0$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx\;\longrightarrow\;\frac{(a+b\,x)^{m+1}\,(c+d\,x)^n}{b\,(m+n+1)}+\frac{n\,(b\,c-a\,d)}{b\,(m+n+1)}\int (a+b\,x)^m\,(c+d\,x)^{n-1}\,dx$$

Program code:

```
Int[1/((a_+b_.*x_)^(5/4)*(c_+d_.*x_)^(1/4)),x_Symbol] :=
  -2/(b*(a+b*x)^(1/4)*(c+d*x)^(1/4)) + c*Int[1/((a+b*x)^(5/4)*(c+d*x)^(5/4)),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && NegQ[a^2*b^2]
```

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n/(b*(m+n+1)) +
  2*c*n/(m+n+1)*Int[(a+b*x)^m*(c+d*x)^(n-1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[b*c+a*d,0] && IGtQ[m+1/2,0] && IGtQ[n+1/2,0] && LtQ[m,n]
```

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^n/(b*(m+n+1)) +
  n*(b*c-a*d)/(b*(m+n+1))*Int[(a+b*x)^m*(c+d*x)^(n-1),x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && GtQ[n,0] && NeQ[m+n+1,0] &&
  Not[IGtQ[m,0] && (Not[IntegerQ[n]] || GtQ[m,0] && LtQ[m-n,0])] &&
  Not[ILtQ[m+n+2,0]] && IntLinearQ[a,b,c,d,m,n,x]
```

6: $\int (a+b\,x)^m\,(c+d\,x)^n\,dx$ when $b\,c-a\,d\neq 0\,\wedge\,m<-1$

Reference: G&R 2.155, CRC 59a

Reference: G&R 2.110.2 or 2.110.6 with $k=1$

Derivation: Linear recurrence 3

Derivation: Integration by parts

Basis: $(a+b\,x)^m\,(c+d\,x)^n == (c+d\,x)^{m+n+2}\,\frac{(a+b\,x)^m}{(c+d\,x)^{m+2}}$

Rule 1.1.1.2.6: If $b\,c-a\,d\neq 0\,\wedge\,m<-1$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx \longrightarrow \frac{(a+b\,x)^{m+1}\,(c+d\,x)^{n+1}}{(b\,c-a\,d)\,(m+1)}-\frac{d\,(m+n+2)}{(b\,c-a\,d)\,(m+1)}\int (a+b\,x)^{m+1}\,(c+d\,x)^n\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)*(c+d*x)^(n+1)/((b*c-a*d)*(m+1)) -
  d*(m+n+2)/((b*c-a*d)*(m+1))*Int[(a+b*x)^(m+1)*(c+d*x)^n,x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[b*c-a*d,0] && LtQ[m,-1] &&
  Not[LtQ[n,-1] && (EqQ[a,0] || NeQ[c,0] && LtQ[m-n,0] && IntegerQ[n])] && IntLinearQ[a,b,c,d,m,n,x]
```

7. $\int (a+b\,x)^m\,(c+d\,x)^n\,dx$ when $b\,c-a\,d\neq 0\land -1\leq m<0\land -1<n<0$

1. $\int \frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}}\,dx$ when $a+c==0\land b-d==0\land a>0$

Rule 1.1.1.2.7.1.1: If $a+c==0\land b-d==0\land a>0$, then

$$\int \frac{1}{\sqrt{a+b\,x}\,\sqrt{c+d\,x}}\,dx\longrightarrow \frac{1}{b}\,\text{ArcCosh}\left[\frac{b\,x}{a}\right]$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_+d_.*x_]),x_Symbol] :=
  ArcCosh[b*x/a]/b /;
FreeQ[{a,b,c,d},x] && EqQ[a+c,0] && EqQ[b-d,0] && GtQ[a,0]
```

2: $\int \frac{1}{\sqrt{a+b\,x}\;\sqrt{c+d\,x}}\,dx$ when $b+d==0\ \land\ a+c>0$

Derivation: Algebraic simplification

Basis: If $a+c>0$, then $(a+b\,x)^m\,(c-b\,x)^m == ((a+b\,x)\,(c-b\,x))^m == \left(a\,c-b\,(a-c)\,x-b^2\,x^2\right)^m$

Rule 1.1.1.2.7.1.2: If $b+d==0\ \land\ a+c>0$, then

$$\int \frac{1}{\sqrt{a+b\,x}\;\sqrt{c+d\,x}}\,dx \longrightarrow \int \frac{1}{\sqrt{a\,c-b\,(a-c)\,x-b^2\,x^2}}\,dx$$

Program code:

```
Int[1/(Sqrt[a_+b_.*x_]*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  Int[1/Sqrt[a*c-b*(a-c)*x-b^2*x^2],x] /;
FreeQ[{a,b,c,d},x] && EqQ[b+d,0] && GtQ[a+c,0]
```

3: $\int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}}\,dx$ when $b c-a d>0 \land b>0$

Derivation: Integration by substitution

Basis: If $b>0$, then $\frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}} == \frac{2}{\sqrt{b}}\,\text{Subst}\left[\frac{1}{\sqrt{b c-a d+d x^2}},\,x,\,\sqrt{a+b x}\right]\partial_x\sqrt{a+b x}$

Rule 1.1.1.2.7.1.3: If $b c-a d>0 \land b>0$, then

$$\int \frac{1}{\sqrt{a+b x}\,\sqrt{c+d x}}\,dx \to \frac{2}{\sqrt{b}}\,\text{Subst}\left[\int \frac{1}{\sqrt{b c-a d+d x^2}}\,dx,\,x,\,\sqrt{a+b x}\right]$$

Program code:

```
Int[1/(Sqrt[a_.+b_.*x_]*Sqrt[c_.+d_.*x_]),x_Symbol] :=
  2/Sqrt[b]*Subst[Int[1/Sqrt[b*c-a*d+d*x^2],x],x,Sqrt[a+b*x]] /;
FreeQ[{a,b,c,d},x] && GtQ[b*c-a*d,0] && GtQ[b,0]
```

2. $\int \frac{1}{(a+b x)\,(c+d x)^{1/3}}\,dx$ when $b c-a d\neq 0$

1: $\int \frac{1}{(a+b x)\,(c+d x)^{1/3}}\,dx$ when $\frac{b c-a d}{b}>0$

Derivation: Integration by substitution

Basis: Let $q=\left(\frac{b c-a d}{b}\right)^{1/3}$, then $\frac{1}{(a+b x)\,(c+d x)^{1/3}} == -\frac{1}{2 q\,(a+b x)}-\text{Subst}\left[\frac{3}{2 b q\,(q-x)}-\frac{3}{2 b\,(q^2+q x+x^2)},\,x,\,(c+d x)^{1/3}\right]\partial_x\,(c+d x)^{1/3}$

Rule 1.1.1.2.7.2.1: If $\frac{b c-a d}{b}>0$, let $q=\left(\frac{b c-a d}{b}\right)^{1/3}$, then

$$\int \frac{1}{(a+b x)\,(c+d x)^{1/3}}\,dx \to$$

$$-\frac{\text{Log}[a+b\,x]}{2\,b\,q}-\frac{3}{2\,b\,q}\,\text{Subst}\left[\int\frac{1}{q-x}\,dx,\,x,\,(c+d\,x)^{1/3}\right]+\frac{3}{2\,b}\,\text{Subst}\left[\int\frac{1}{q^2+q\,x+x^2}\,dx,\,x,\,(c+d\,x)^{1/3}\right]$$

Program code:

```
Int[1/((a_.+b_.*x_)*(c_.+d_.*x_)^(1/3)),x_Symbol] :=
  With[{q=Rt[(b*c-a*d)/b,3]},
  -Log[RemoveContent[a+b*x,x]]/(2*b*q) -
  3/(2*b*q)*Subst[Int[1/(q-x),x],x,(c+d*x)^(1/3)] +
  3/(2*b)*Subst[Int[1/(q^2+q*x+x^2),x],x,(c+d*x)^(1/3)]] /;
FreeQ[{a,b,c,d},x] && PosQ[(b*c-a*d)/b]
```

2: $\int\frac{1}{(a+b\,x)\,(c+d\,x)^{1/3}}\,dx$ when $\frac{b\,c-a\,d}{b}\ngtr 0$

Derivation: Integration by substitution

Basis: Let $q=\left(-\frac{b\,c-a\,d}{b}\right)^{1/3}$, then $\frac{1}{(a+b\,x)\,(c+d\,x)^{1/3}}=\frac{1}{2\,q\,(a+b\,x)}-\text{Subst}\left[\frac{3}{2\,b\,q\,(q+x)}-\frac{3}{2\,b\,(q^2-q\,x+x^2)},\,x,\,(c+d\,x)^{1/3}\right]\partial_x(c+d\,x)^{1/3}$

Rule 1.1.1.2.7.2.2: If $\frac{b\,c-a\,d}{b}\ngtr 0$, let $q=\left(-\frac{b\,c-a\,d}{b}\right)^{1/3}$, then

$$\int\frac{1}{(a+b\,x)\,(c+d\,x)^{1/3}}\,dx\,\longrightarrow$$

$$\frac{\text{Log}[a+b\,x]}{2\,b\,q}-\frac{3}{2\,b\,q}\,\text{Subst}\left[\int\frac{1}{q+x}\,dx,\,x,\,(c+d\,x)^{1/3}\right]+\frac{3}{2\,b}\,\text{Subst}\left[\int\frac{1}{q^2-q\,x+x^2}\,dx,\,x,\,(c+d\,x)^{1/3}\right]$$

Program code:

```
Int[1/((a_.+b_.*x_)*(c_.+d_.*x_)^(1/3)),x_Symbol] :=
  With[{q=Rt[-(b*c-a*d)/b,3]},
  Log[RemoveContent[a+b*x,x]]/(2*b*q) -
  3/(2*b*q)*Subst[Int[1/(q+x),x],x,(c+d*x)^(1/3)] +
  3/(2*b)*Subst[Int[1/(q^2-q*x+x^2),x],x,(c+d*x)^(1/3)]] /;
FreeQ[{a,b,c,d},x] && NegQ[(b*c-a*d)/b]
```

3. $\int \frac{1}{(a+b\,x)\,(c+d\,x)^{2/3}}\,dx$ when $b\,c-a\,d\neq 0$

1: $\int \frac{1}{(a+b\,x)\,(c+d\,x)^{2/3}}\,dx$ when $\frac{b\,c-a\,d}{b}>0$

Derivation: Integration by substitution

Basis: Let $q=\left(\frac{b\,c-a\,d}{b}\right)^{1/3}$, then $\frac{1}{(a+b\,x)\,(c+d\,x)^{2/3}}==-\frac{1}{2\,q^2\,(a+b\,x)}-\text{Subst}\left[\frac{3}{2\,b\,q^2\,(q-x)}+\frac{3}{2\,b\,q\,(q^2+q\,x+x^2)},\,x,\,(c+d\,x)^{1/3}\right]\partial_x\,(c+d\,x)^{1/3}$

Rule 1.1.1.2.7.3.1: If $\frac{b\,c-a\,d}{b}>0$, let $q=\left(\frac{b\,c-a\,d}{b}\right)^{1/3}$, then

$$\int \frac{1}{(a+b\,x)\,(c+d\,x)^{2/3}}\,dx\;\longrightarrow$$

$$-\frac{\text{Log}[a+b\,x]}{2\,b\,q^2}-\frac{3}{2\,b\,q^2}\,\text{Subst}\left[\int\frac{1}{q-x}\,dx,\,x,\,(c+d\,x)^{1/3}\right]-\frac{3}{2\,b\,q}\,\text{Subst}\left[\int\frac{1}{q^2+q\,x+x^2}\,dx,\,x,\,(c+d\,x)^{1/3}\right]$$

Program code:

```
Int[1/((a_.+b_.*x_)*(c_.+d_.*x_)^(2/3)),x_Symbol] :=
  With[{q=Rt[(b*c-a*d)/b,3]},
  -Log[RemoveContent[a+b*x,x]]/(2*b*q^2) -
   3/(2*b*q^2)*Subst[Int[1/(q-x),x],x,(c+d*x)^(1/3)] -
  3/(2*b*q)*Subst[Int[1/(q^2+q*x+x^2),x],x,(c+d*x)^(1/3)]] /;
FreeQ[{a,b,c,d},x] && PosQ[(b*c-a*d)/b]
```

2: $\int \frac{1}{(a+b x)\ (c+d x)^{2/3}}\,dx$ when $\frac{b c-a d}{b} \not> 0$

Derivation: Integration by substitution

Basis: Let $q = \left(-\frac{b c-a d}{b}\right)^{1/3}$, then $\frac{1}{(a+b x)\ (c+d x)^{2/3}} == -\frac{1}{2 q^2\ (a+b x)} + \text{Subst}\left[\frac{3}{2 b q^2\ (q+x)} + \frac{3}{2 b q\ (q^2-q x+x^2)},\ x,\ (c+d x)^{1/3}\right] \partial_x (c+d x)^{1/3}$

Rule 1.1.1.2.7.3.2: If $\frac{b c-a d}{b} \not> 0$, let $q = \left(-\frac{b c-a d}{b}\right)^{1/3}$, then

$$\int \frac{1}{(a+b x)\ (c+d x)^{2/3}}\,dx \longrightarrow$$

$$-\frac{\text{Log}[a+b x]}{2 b q^2} + \frac{3}{2 b q^2}\,\text{Subst}\left[\int \frac{1}{q+x}\,dx,\ x,\ (c+d x)^{1/3}\right] + \frac{3}{2 b q}\,\text{Subst}\left[\int \frac{1}{q^2-q x+x^2}\,dx,\ x,\ (c+d x)^{1/3}\right]$$

Program code:

```
Int[1/((a_.+b_.*x_)*(c_.+d_.*x_)^(2/3)),x_Symbol] :=
  With[{q=Rt[-(b*c-a*d)/b,3]},
  -Log[RemoveContent[a+b*x,x]]/(2*b*q^2) +
   3/(2*b*q^2)*Subst[Int[1/(q+x),x],x,(c+d*x)^(1/3)] +
  3/(2*b*q)*Subst[Int[1/(q^2-q*x+x^2),x],x,(c+d*x)^(1/3)]] /;
FreeQ[{a,b,c,d},x] && NegQ[(b*c-a*d)/b]
```

4. $\int \frac{1}{(a+b\,x)^{1/3}\,(c+d\,x)^{2/3}}\,dx$ when $b\,c-a\,d\neq 0$

   1: $\int \frac{1}{(a+b\,x)^{1/3}\,(c+d\,x)^{2/3}}\,dx$ when $b\,c-a\,d\neq 0\,\land\,\frac{d}{b}>0$

Rule 1.1.1.2.7.4.1: If $b\,c-a\,d\neq 0\,\land\,\frac{d}{b}>0$, let $q=\left(\frac{d}{b}\right)^{1/3}$, then

$$\int \frac{1}{(a+b\,x)^{1/3}\,(c+d\,x)^{2/3}}\,dx \longrightarrow -\frac{\sqrt{3}\,q}{d}\,\text{ArcTan}\Big[\frac{2\,q\,(a+b\,x)^{1/3}}{\sqrt{3}\,(c+d\,x)^{1/3}}+\frac{1}{\sqrt{3}}\Big]-\frac{q}{2\,d}\,\text{Log}[c+d\,x]-\frac{3\,q}{2\,d}\,\text{Log}\Big[\frac{q\,(a+b\,x)^{1/3}}{(c+d\,x)^{1/3}}-1\Big]$$

Program code:

```
Int[1/((a_.+b_.*x_)^(1/3)*(c_.+d_.*x_)^(2/3)),x_Symbol] :=
  With[{q=Rt[d/b,3]},
  -Sqrt[3]*q/d*ArcTan[2*q*(a+b*x)^(1/3)/(Sqrt[3]*(c+d*x)^(1/3))+1/Sqrt[3]] -
  q/(2*d)*Log[c+d*x] -
  3*q/(2*d)*Log[q*(a+b*x)^(1/3)/(c+d*x)^(1/3)-1]] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && PosQ[d/b]
```

2: $\int \frac{1}{(a+b\,x)^{1/3}\,(c+d\,x)^{2/3}}\,dx$ when $b\,c-a\,d\neq 0\ \land\ \frac{d}{b}\not>0$

Rule 1.1.1.2.7.4.2: If $b\,c-a\,d\neq 0\ \land\ \frac{d}{b}\not>0$, let $q=\left(-\frac{d}{b}\right)^{1/3}$, then

$$\int \frac{1}{(a+b\,x)^{1/3}\,(c+d\,x)^{2/3}}\,dx \longrightarrow \frac{\sqrt{3}\,q}{d}\,\text{ArcTan}\!\left[\frac{1}{\sqrt{3}}-\frac{2\,q\,(a+b\,x)^{1/3}}{\sqrt{3}\,(c+d\,x)^{1/3}}\right]+\frac{q}{2\,d}\,\text{Log}[c+d\,x]+\frac{3\,q}{2\,d}\,\text{Log}\!\left[\frac{q\,(a+b\,x)^{1/3}}{(c+d\,x)^{1/3}}+1\right]$$

Program code:

```
Int[1/((a_.+b_.*x_)^(1/3)*(c_.+d_.*x_)^(2/3)),x_Symbol] :=
  With[{q=Rt[-d/b,3]},
  Sqrt[3]*q/d*ArcTan[1/Sqrt[3]-2*q*(a+b*x)^(1/3)/(Sqrt[3]*(c+d*x)^(1/3))] +
   q/(2*d)*Log[c+d*x] +
  3*q/(2*d)*Log[q*(a+b*x)^(1/3)/(c+d*x)^(1/3)+1]] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && NegQ[d/b]
```

5: $\int (a+b\,x)^m\,(c+d\,x)^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,-1<m<0\,\land\,n==m\,\land\,3\leq \text{Denominator}[m]\leq 4$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a+b\,x)^m\,(c+d\,x)^m}{((a+b\,x)\,(c+d\,x))^m}==0$

Rule 1.1.1.2.7.5: If $b\,c-a\,d\neq 0\,\land\,-1<m<0\,\land\,3\leq \text{Denominator}[m]\leq 4$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx\,\longrightarrow\,\frac{(a+b\,x)^m\,(c+d\,x)^m}{\left(a\,c+(b\,c+a\,d)\,x+b\,d\,x^2\right)^m}\int \left(a\,c+(b\,c+a\,d)\,x+b\,d\,x^2\right)^m\,dx$$

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx\,\longrightarrow\,\frac{(a+b\,x)^m\,(c+d\,x)^m}{((a+b\,x)\,(c+d\,x))^m}\int \left(a\,c+(b\,c+a\,d)\,x+b\,d\,x^2\right)^m\,dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_+d_.*x_)^m_,x_Symbol] :=
  (a+b*x)^m*(c+d*x)^m/(a*c+(b*c+a*d)*x+b*d*x^2)^m*Int[(a*c+(b*c+a*d)*x+b*d*x^2)^m,x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && LtQ[-1,m,0] && LeQ[3,Denominator[m],4] && AtomQ[b*c+a*d]
```

```
Int[(a_.+b_.*x_)^m_*(c_+d_.*x_)^m_,x_Symbol] :=
  (a+b*x)^m*(c+d*x)^m/((a+b*x)*(c+d*x))^m*Int[(a*c+(b*c+a*d)*x+b*d*x^2)^m,x] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && LtQ[-1,m,0] && LeQ[3,Denominator[m],4]
```

6: $\int (a+b\,x)^m\,(c+d\,x)^n\,dx$ when $b\,c-a\,d\neq 0\,\land\,-1<m<0\,\land\,-1\leq n<0$

Derivation: Integration by substitution

Basis: If $p\in\mathbb{Z}^+$, then

$(a+b\,x)^m\,(c+d\,x)^n == \frac{p}{b}\,\text{Subst}\left[x^{p\,(m+1)-1}\,\left(c-\frac{a\,d}{b}+\frac{d}{b}\,x^p\right)^n,\,x,\,(a+b\,x)^{1/p}\right]\,\partial_x\,(a+b\,x)^{1/p}$

Rule 1.1.1.2.7.7: If $b\,c-a\,d\neq 0\,\land\,-1<m<0\,\land\,-1\leq n<0$, let $p=\text{Denominator}[m]$, then

$$\int (a+b\,x)^m\,(c+d\,x)^n\,dx\;\longrightarrow\;\frac{p}{b}\,\text{Subst}\left[\int x^{p\,(m+1)-1}\,\left(c-\frac{a\,d}{b}+\frac{d\,x^p}{b}\right)^n\,dx,\,x,\,(a+b\,x)^{1/p}\right]$$

Program code:

```
Int[(a_.+b_.*x_)^m_*(c_.+d_.*x_)^n_,x_Symbol] :=
  With[{p=Denominator[m]},
  p/b*Subst[Int[x^(p*(m+1)-1)*(c-a*d/b+d*x^p/b)^n,x],x,(a+b*x)^(1/p)]] /;
FreeQ[{a,b,c,d},x] && NeQ[b*c-a*d,0] && LtQ[-1,m,0] && LeQ[-1,n,0] && LeQ[Denominator[n],Denominator[m]] &&
  IntLinearQ[a,b,c,d,m,n,x]
```

H. $\int (a+b\,x)^m\,(c+d\,x)^n\,dx$ when $b\,c-a\,d\neq 0$

1. $\int (b\,x)^m\,(c+d\,x)^n\,dx$

1: $\int (b\,x)^m\,(c+d\,x)^n\,dx$ when $m\notin\mathbb{Z}\land(n\in\mathbb{Z}\lor c>0)$

Rule 1.1.1.2.H.1.1: If $m\notin\mathbb{Z}\land(n\in\mathbb{Z}\lor c>0)$, then

$$\int (b\,x)^m\,(c+d\,x)^n\,dx\to\frac{c^n\,(b\,x)^{m+1}}{b\,(m+1)}\,\text{Hypergeometric2F1}\left[-n,\,m+1,\,m+2,\,-\frac{d\,x}{c}\right]$$

Program code:

```
Int[(b_.*x_)^m_*(c_+d_.*x_)^n_,x_Symbol] :=
  c^n*(b*x)^(m+1)/(b*(m+1))*Hypergeometric2F1[-n,m+1,m+2,-d*x/c] /;
FreeQ[{b,c,d,m,n},x] && Not[IntegerQ[m]] && (IntegerQ[n] || GtQ[c,0] && Not[EqQ[n,-1/2] && EqQ[c^2-d^2,0] && GtQ[-d/(b*c),0]])
```

2: $\int (b\,x)^m\,(c+d\,x)^n\,dx$ when $n\notin\mathbb{Z}\land\left(m\in\mathbb{Z}\lor-\frac{d}{b\,c}>0\right)$

Rule 1.1.1.2.H.1.2: If $n\notin\mathbb{Z}\land\left(m\in\mathbb{Z}\lor-\frac{d}{b\,c}>0\right)$, then

$$\int (b\,x)^m\,(c+d\,x)^n\,dx\to\frac{(c+d\,x)^{n+1}}{d\,(n+1)\,\left(-\frac{d}{b\,c}\right)^m}\,\text{Hypergeometric2F1}\left[-m,\,n+1,\,n+2,\,1+\frac{d\,x}{c}\right]$$

Program code:

```
Int[(b_.*x_)^m_*(c_+d_.*x_)^n_,x_Symbol] :=
  (c+d*x)^(n+1)/(d*(n+1)*(-d/(b*c))^m)*Hypergeometric2F1[-m,n+1,n+2,1+d*x/c] /;
FreeQ[{b,c,d,m,n},x] && Not[IntegerQ[n]] && (IntegerQ[m] || GtQ[-d/(b*c),0])
```

3. $\int (b\,x)^m\,(c+d\,x)^n\,dx$ when $m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,c\not>0\,\land\,-\frac{d}{b\,c}\not>0$

1: $\int (b\,x)^m\,(c+d\,x)^n\,dx$ when $m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,c\not>0\,\land\,-\frac{d}{b\,c}\not>0\,\land\,(m\in\mathbb{R}\,\lor\,n\notin\mathbb{R})$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c+d\,x)^n}{\left(1+\frac{d\,x}{c}\right)^n} == 0$

Rule 1.1.1.2.H.1.3.1: If $m\notin\mathbb{Z}\,\land\,n\notin\mathbb{Z}\,\land\,c\not>0\,\land\,-\frac{d}{b\,c}\not>0\,\land\,(m\in\mathbb{R}\,\lor\,n\notin\mathbb{R})$, then

$$\int (b\,x)^m\,(c+d\,x)^n\,dx\,\longrightarrow\,\frac{c^{\text{IntPart}[n]}\,(c+d\,x)^{\text{FracPart}[n]}}{\left(1+\frac{d\,x}{c}\right)^{\text{FracPart}[n]}}\int (b\,x)^m\left(1+\frac{d\,x}{c}\right)^n dx$$

Program code:

```
Int[(b_.*x_)^m_*(c_+d_.*x_)^n_,x_Symbol] :=
  c^IntPart[n]*(c+d*x)^FracPart[n]/(1+d*x/c)^FracPart[n]*Int[(b*x)^m*(1+d*x/c)^n,x] /;
FreeQ[{b,c,d,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && Not[GtQ[c,0]] && Not[GtQ[-d/(b*c),0]] &&
  (RationalQ[m] && Not[EqQ[n,-1/2] && EqQ[c^2-d^2,0]] || Not[RationalQ[n]])
```

2: $\int (b\,x)^m\,(c+d\,x)^n\,dx$ when $m\notin\mathbb{Z}\land n\notin\mathbb{Z}\land c\ngtr 0\land -\frac{d}{b\,c}\ngtr 0\land\neg\,(m\in\mathbb{R}\lor n\notin\mathbb{R})$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(b\,x)^m}{\left(-\frac{d\,x}{c}\right)^m} == 0$

Rule 1.1.1.2.H.1.3.2: If $m\notin\mathbb{Z}\land n\notin\mathbb{Z}\land c\ngtr 0\land -\frac{d}{b\,c}\ngtr 0\land\neg\,(m\in\mathbb{R}\lor n\notin\mathbb{R})$, then

$$\int (b\,x)^m\,(c+d\,x)^n\,dx\;\longrightarrow\;\frac{\left(-\frac{b\,c}{d}\right)^{\text{IntPart}[m]}\,(b\,x)^{\text{FracPart}[m]}}{\left(-\frac{d\,x}{c}\right)^{\text{FracPart}[m]}}\int\left(-\frac{d\,x}{c}\right)^m\,(c+d\,x)^n\,dx$$

Program code:

```
Int[(b_.*x_)^m_*(c_+d_.*x_)^n_,x_Symbol] :=
  (-b*c/d)^IntPart[m]*(b*x)^FracPart[m]/(-d*x/c)^FracPart[m]*Int[(-d*x/c)^m*(c+d*x)^n,x] /;
FreeQ[{b,c,d,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && Not[GtQ[c,0]] && Not[GtQ[-d/(b*c),0]]
```

2. $\int (a + b x)^m (c + d x)^n \, dx$ when $b c - a d \neq 0 \land m \notin \mathbb{Z}$

1: $\int (a + b x)^m (c + d x)^n \, dx$ when $b c - a d \neq 0 \land m \notin \mathbb{Z} \land \left(n \in \mathbb{Z} \lor \frac{b}{b c - a d} > 0\right)$

Rule 1.1.1.2.H.2.2.1: If $b c - a d \neq 0 \land m \notin \mathbb{Z} \land \left(n \in \mathbb{Z} \lor \frac{b}{b c - a d} > 0\right)$, then

$$\int (a + b x)^m (c + d x)^n \, dx \rightarrow \frac{(a + b x)^{m+1}}{b (m + 1) \left(\frac{b}{b c - a d}\right)^n} \text{Hypergeometric2F1}\left[-n, m + 1, m + 2, -\frac{d (a + b x)}{b c - a d}\right]$$

Program code:

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^n_,x_Symbol] :=
  (b*c-a*d)^n*(a+b*x)^(m+1)/(b^(n+1)*(m+1))*Hypergeometric2F1[-n,m+1,m+2,-d*(a+b*x)/(b*c-a*d)] /;
FreeQ[{a,b,c,d,m},x] && NeQ[b*c-a*d,0] && Not[IntegerQ[m]] && IntegerQ[n]
```

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^n_,x_Symbol] :=
  (a+b*x)^(m+1)/(b*(m+1)*(b/(b*c-a*d))^n)*Hypergeometric2F1[-n,m+1,m+2,-d*(a+b*x)/(b*c-a*d)] /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[b*c-a*d,0] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && GtQ[b/(b*c-a*d),0] &&
  (RationalQ[m] || Not[RationalQ[n] && GtQ[-d/(b*c-a*d),0]])
```

2: $\int (a+b x)^m (c+d x)^n \, dx$ when $b c-a d \neq 0 \land m \notin \mathbb{Z} \land n \notin \mathbb{Z} \land \frac{b}{b c-a d} \ngtr 0$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(c+d x)^n}{\left(\frac{b c}{b c-a d}+\frac{b d x}{b c-a d}\right)^n} == 0$

Rule 1.1.1.2.H.2.2.2: If $b c-a d \neq 0 \land m \notin \mathbb{Z} \land n \notin \mathbb{Z} \land \frac{b}{b c-a d} \ngtr 0$, then

$$\int (a+b x)^m (c+d x)^n \, dx \to \frac{(c+d x)^{\text{FracPart}[n]}}{\left(\frac{b}{b c-a d}\right)^{\text{IntPart}[n]} \left(\frac{b (c+d x)}{b c-a d}\right)^{\text{FracPart}[n]}} \int (a+b x)^m \left(\frac{b c}{b c-a d}+\frac{b d x}{b c-a d}\right)^n dx$$

Program code:

```
Int[(a_+b_.*x_)^m_*(c_+d_.*x_)^n_,x_Symbol] :=
  (c+d*x)^FracPart[n]/((b/(b*c-a*d))^IntPart[n]*(b*(c+d*x)/(b*c-a*d))^FracPart[n])*
    Int[(a+b*x)^m*Simp[b*c/(b*c-a*d)+b*d*x/(b*c-a*d),x]^n,x] /;
FreeQ[{a,b,c,d,m,n},x] && NeQ[b*c-a*d,0] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && (RationalQ[m] || Not[SimplerQ[n+1,m+1]])
```

S: $\int (a+b\,u)^m\,(c+d\,u)^n\,dx$ when $u == e+f\,x$

Derivation: Integration by substitution

Rule 1.1.1.2.S: If $u == e+f\,x$, then

$$\int (a+b\,u)^m\,(c+d\,u)^n\,dx \longrightarrow \frac{1}{f}\,\text{Subst}\left[\int (a+b\,x)^m\,(c+d\,x)^n\,dx,\,x,\,u\right]$$

Program code:

```
Int[(a_.+b_.*u_)^m_.*(c_.+d_.*u_)^n_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[(a+b*x)^m*(c+d*x)^n,x],x,u] /;
FreeQ[{a,b,c,d,m,n},x] && LinearQ[u,x] && NeQ[Coefficient[u,x,0],0]
```

```
(* IntLinearQ[a,b,c,d,m,n,x] returns True iff (a+b*x)^m*(c+d*x)^n is integrable wrt x in terms of non-hypergeometric functions. *)
IntLinearQ[a_,b_,c_,d_,m_,n_,x_] :=
  IGtQ[m,0] || IGtQ[n,0] || IntegersQ[3*m,3*n] || IntegersQ[4*m,4*n] || IntegersQ[2*m,6*n] || IntegersQ[6*m,2*n] || ILtQ[m+n,-1] || In
```