Rules for integrands involving logarithms

1. $\int u \frac{\text{Log}[1 - F[x]]\, F'[x]}{F[x]}\, dx$

1: $\int \frac{\text{Log}[1 - F[x]]\, F'[x]}{F[x]}\, dx$

Basis: $\partial_x \text{PolyLog}[2, x] == \frac{\text{PolyLog}[1,x]}{x} == -\frac{\text{Log}[1-x]}{x}$

Rule:

$$\int \frac{\text{Log}[1 - F[x]]\, F'[x]}{F[x]}\, dx \longrightarrow -\text{PolyLog}[2, F[x]]$$

Program code:

```
Int[u_*Log[v_],x_Symbol] :=
  With[{w=DerivativeDivides[v,u*(1-v),x]},
  w*PolyLog[2,1-v] /;
 Not[FalseQ[w]]]
```

2: $\int (a+b\,\mathrm{Log}[u])\,\frac{\mathrm{Log}[1-F[x]]\,F'[x]}{F[x]}\,dx$ when u is free of inverse functions

Derivation: Integration by parts

Basis: $\frac{\mathrm{Log}[1-x]}{x} == -\partial_x \mathrm{PolyLog}[2,\,x]$

Rule:

$$\int (a+b\,\mathrm{Log}[u])\,\frac{\mathrm{Log}[1-F[x]]\,F'[x]}{F[x]}\,dx \longrightarrow -(a+b\,\mathrm{Log}[u])\,\mathrm{PolyLog}[2,\,F[x]]+b\int \frac{\mathrm{PolyLog}[2,\,F[x]]\,\partial_x u}{u}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[u_])*Log[v_]*w_,x_Symbol] :=
  With[{z=DerivativeDivides[v,w*(1-v),x]},
  z*(a+b*Log[u])*PolyLog[2,1-v] -
  b*Int[SimplifyIntegrand[z*PolyLog[2,1-v]*D[u,x]/u,x],x] /;
 Not[FalseQ[z]]] /;
FreeQ[{a,b},x] && InverseFunctionFreeQ[u,x]
```

2. $\int u\,(a + b\,\text{Log}[c\,\text{Log}[d\,x^n]^p])\,dx$

1: $\int \text{Log}[c\,\text{Log}[d\,x^n]^p]\,dx$

Derivation: Integration by parts

Basis: $\partial_x\,\text{Log}[c\,\text{Log}[d\,x^n]^p] == \frac{n\,p}{x\,\text{Log}[d\,x^n]}$

Rule:

$$\int \text{Log}[c\,\text{Log}[d\,x^n]^p]\,dx \longrightarrow x\,\text{Log}[c\,\text{Log}[d\,x^n]^p] - n\,p\int \frac{1}{\text{Log}[d\,x^n]}\,dx$$

Program code:

```
Int[Log[c_.*Log[d_.*x_^n_.]^p_.],x_Symbol] :=
  x*Log[c*Log[d*x^n]^p] - n*p*Int[1/Log[d*x^n],x] /;
FreeQ[{c,d,n,p},x]
```

2. $\int (e\,x)^m\,(a + b\,\text{Log}[c\,\text{Log}[d\,x^n]^p])\,dx$

1: $\int \frac{a + b\,\text{Log}[c\,\text{Log}[d\,x^n]^p]}{x}\,dx$

Derivation: Integration by parts

Basis: $\frac{1}{x} == \partial_x\,\frac{\text{Log}[d\,x^n]}{n}$

Basis: $\partial_x\,(a + b\,\text{Log}[c\,\text{Log}[d\,x^n]^p]) == \frac{b\,n\,p}{x\,\text{Log}[d\,x^n]}$

Rule:

$$\int \frac{a+b\,\text{Log}\left[c\,\text{Log}\left[d\,x^n\right]^p\right]}{x}\,dx \longrightarrow \frac{\text{Log}\left[d\,x^n\right]\,\left(a+b\,\text{Log}\left[c\,\text{Log}\left[d\,x^n\right]^p\right]\right)}{n} - b\,p\int \frac{1}{x}\,dx \longrightarrow \frac{\text{Log}\left[d\,x^n\right]\,\left(a+b\,\text{Log}\left[c\,\text{Log}\left[d\,x^n\right]^p\right]\right)}{n} - b\,p\,\text{Log}[x]$$

Program code:

```
Int[(a_.+b_.*Log[c_.*Log[d_.*x_^n_.]^p_.])/x_,x_Symbol] :=
  Log[d*x^n]*(a+b*Log[c*Log[d*x^n]^p])/n - b*p*Log[x] /;
FreeQ[{a,b,c,d,n,p},x]
```

2: $\int (e\,x)^m\,\left(a+b\,\text{Log}\left[c\,\text{Log}\left[d\,x^n\right]^p\right]\right)\,dx$ when $m \neq -1$

Derivation: Integration by parts

Basis: $\partial_x\,\left(a+b\,\text{Log}\left[c\,\text{Log}\left[d\,x^n\right]^p\right]\right) == \frac{b\,n\,p}{x\,\text{Log}\left[d\,x^n\right]}$

Rule: If $m \neq -1$, then

$$\int (e\,x)^m\,\left(a+b\,\text{Log}\left[c\,\text{Log}\left[d\,x^n\right]^p\right]\right)\,dx \longrightarrow \frac{(e\,x)^{m+1}\,\left(a+b\,\text{Log}\left[c\,\text{Log}\left[d\,x^n\right]^p\right]\right)}{e\,(m+1)} - \frac{b\,n\,p}{m+1}\int \frac{(e\,x)^m}{\text{Log}\left[d\,x^n\right]}\,dx$$

Program code:

```
Int[(e_.*x_)^m_.*(a_.+b_.*Log[c_.*Log[d_.*x_^n_.]^p_.]),x_Symbol] :=
  (e*x)^(m+1)*(a+b*Log[c*Log[d*x^n]^p])/(e*(m+1)) - b*n*p/(m+1)*Int[(e*x)^m/Log[d*x^n],x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && NeQ[m,-1]
```

3. $\int u\,(a+b\,\mathrm{Log}[c\,RF_x{}^p])^n\,dx$ when $n\in\mathbb{Z}^+$

1: $\int (a+b\,\mathrm{Log}[c\,RF_x{}^p])^n\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Integration by parts

Basis: $\partial_x\,(a+b\,\mathrm{Log}[c\,RF_x{}^p])^n == \frac{b\,n\,p\,(a+b\,\mathrm{Log}[c\,RF_x{}^p])^{n-1}\,\partial_x RF_x}{RF_x}$

Rule: If $n\in\mathbb{Z}^+$, then

$$\int (a+b\,\mathrm{Log}[c\,RF_x{}^p])^n\,dx \,\longrightarrow\, x\,(a+b\,\mathrm{Log}[c\,RF_x{}^p])^n - b\,n\,p\int\frac{x\,(a+b\,\mathrm{Log}[c\,RF_x{}^p])^{n-1}\,\partial_x RF_x}{RF_x}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*RFx_^p_.])^n_.,x_Symbol] :=
  x*(a+b*Log[c*RFx^p])^n -
  b*n*p*Int[SimplifyIntegrand[x*(a+b*Log[c*RFx^p])^(n-1)*D[RFx,x]/RFx,x],x] /;
FreeQ[{a,b,c,p},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

2. $\int (d+e\,x)^m\,(a+b\,\mathrm{Log}[c\,RF_x{}^p])^n\,dx$ when $n\in\mathbb{Z}^+\land(n==1\lor m\in\mathbb{Z})$

1: $\int \frac{(a+b\,\mathrm{Log}[c\,RF_x{}^p])^n}{d+e\,x}\,dx$ when $n\in\mathbb{Z}^+$?? ?? n>1?

Derivation: Integration by parts

Basis: $\frac{1}{d+e\,x} == \partial_x\,\frac{\mathrm{Log}[d+e\,x]}{e}$

Basis: $\partial_x\,(a+b\,\mathrm{Log}[c\,RF_x{}^p])^n == \frac{b\,n\,p\,(a+b\,\mathrm{Log}[c\,RF_x{}^p])^{n-1}\,\partial_x RF_x}{RF_x}$

Rule: If $n\in\mathbb{Z}^+$, then

$$\int \frac{(a+b\,\text{Log}[c\,RF_x{}^p])^n}{d+e\,x}\,dx \to \frac{\text{Log}[d+e\,x]\,(a+b\,\text{Log}[c\,RF_x{}^p])^n}{e} - \frac{b\,n\,p}{e}\int \frac{\text{Log}[d+e\,x]\,(a+b\,\text{Log}[c\,RF_x{}^p])^{n-1}\,\partial_x RF_x}{RF_x}\,dx$$

Program code:

```
Int[(a_.+b_.*Log[c_.*RFx_^p_.])^n_./(d_.+e_.*x_),x_Symbol] :=
  Log[d+e*x]*(a+b*Log[c*RFx^p])^n/e -
  b*n*p/e*Int[Log[d+e*x]*(a+b*Log[c*RFx^p])^(n-1)*D[RFx,x]/RFx,x] /;
FreeQ[{a,b,c,d,e,p},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0]
```

2: $\int (d+e\,x)^m\,(a+b\,\text{Log}[c\,RF_x{}^p])^n\,dx$ when $n\in\mathbb{Z}^+\land(n==1\lor m\in\mathbb{Z})\land m\neq-1$

Derivation: Integration by parts

Basis: $(d+e\,x)^m == \partial_x \frac{(d+e\,x)^{m+1}}{e\,(m+1)}$

Basis: $\partial_x\,(a+b\,\text{Log}[c\,RF_x{}^p])^n == \frac{b\,n\,p\,(a+b\,\text{Log}[c\,RF_x{}^p])^{n-1}\,\partial_x RF_x}{RF_x}$

Rule: If $n\in\mathbb{Z}^+\land(n==1\lor m\in\mathbb{Z})\land m\neq-1$, then

$$\int (d+e\,x)^m\,(a+b\,\text{Log}[c\,RF_x{}^p])^n\,dx \to \frac{(d+e\,x)^{m+1}\,(a+b\,\text{Log}[c\,RF_x{}^p])^n}{e\,(m+1)} - \frac{b\,n\,p}{e\,(m+1)}\int \frac{(d+e\,x)^{m+1}\,(a+b\,\text{Log}[c\,RF_x{}^p])^{n-1}\,\partial_x RF_x}{RF_x}\,dx$$

Program code:

```
Int[(d_.+e_.*x_)^m_.*(a_.+b_.*Log[c_.*RFx_^p_.])^n_.,x_Symbol] :=
  (d+e*x)^(m+1)*(a+b*Log[c*RFx^p])^n/(e*(m+1)) -
  b*n*p/(e*(m+1))*Int[SimplifyIntegrand[(d+e*x)^(m+1)*(a+b*Log[c*RFx^p])^(n-1)*D[RFx,x]/RFx,x],x] /;
FreeQ[{a,b,c,d,e,m,p},x] && RationalFunctionQ[RFx,x] && IGtQ[n,0] && (EqQ[n,1] || IntegerQ[m]) && NeQ[m,-1]
```

3: $\int \frac{\text{Log}\left[c\, RF_x{}^n\right]}{d+e\,x^2}\,dx$

Derivation: Integration by parts

Rule: Let $u = \int \frac{1}{d+e\,x^2}\,dx$, then

$$\int \frac{\text{Log}\left[c\, RF_x{}^n\right]}{d+e\,x^2}\,dx \longrightarrow u\,\text{Log}\left[c\, RF_x{}^n\right] - n\int \frac{u\,\partial_x RF_x}{RF_x}\,dx$$

Program code:

```
Int[Log[c_.*RFx_^n_.]/(d_+e_.*x_^2),x_Symbol] :=
  With[{u=IntHide[1/(d+e*x^2),x]},
  u*Log[c*RFx^n] - n*Int[SimplifyIntegrand[u*D[RFx,x]/RFx,x],x]] /;
FreeQ[{c,d,e,n},x] && RationalFunctionQ[RFx,x] && Not[PolynomialQ[RFx,x]]
```

4: $\int \frac{\text{Log}[c\, P_x{}^n]}{Q_x}\, dx$ when $\text{QuadraticQ}[Q_x] \wedge \partial_x \frac{P_x}{Q_x} == 0$

Derivation: Integration by parts

Rule: If $\text{QuadraticQ}[Q_x] \wedge \partial_x \frac{P_x}{Q_x} == 0$, let $u = \int \frac{1}{Q_x}\, dx$, then

$$\int \frac{\text{Log}[c\, P_x{}^n]}{Q_x}\, dx \rightarrow u\, \text{Log}[c\, P_x{}^n] - n \int \frac{u\, \partial_x P_x}{P_x}\, dx$$

Program code:

```
Int[Log[c_.*Px_^n_.]/Qx_,x_Symbol] :=
  With[{u=IntHide[1/Qx,x]},
  u*Log[c*Px^n] - n*Int[SimplifyIntegrand[u*D[Px,x]/Px,x],x]] /;
FreeQ[{c,n},x] && QuadraticQ[{Qx,Px},x] && EqQ[D[Px/Qx,x],0]
```

5: $\int RG_x\,(a+b\,\mathrm{Log}[c\,RF_x{}^p])^n\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+$, then

$$\int RG_x\,(a+b\,\mathrm{Log}[c\,RF_x{}^p])^n\,dx \to \int (a+b\,\mathrm{Log}[c\,RF_x{}^p])^n\,\mathrm{ExpandIntegrand}[RG_x,\,x]\,dx$$

Program code:

```
Int[RGx_*(a_.+b_.*Log[c_.*RFx_^p_.])^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[(a+b*Log[c*RFx^p])^n,RGx,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,p},x] && RationalFunctionQ[RFx,x] && RationalFunctionQ[RGx,x] && IGtQ[n,0]
```

```
Int[RGx_*(a_.+b_.*Log[c_.*RFx_^p_.])^n_.,x_Symbol] :=
  With[{u=ExpandIntegrand[RGx*(a+b*Log[c*RFx^p])^n,x]},
  Int[u,x] /;
 SumQ[u]] /;
FreeQ[{a,b,c,p},x] && RationalFunctionQ[RFx,x] && RationalFunctionQ[RGx,x] && IGtQ[n,0]
```

4: $\int RF_x\,(a+b\,\mathrm{Log}[F[(c+d\,x)^{1/n},\,x]])\,dx$ when $n\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $n\in\mathbb{Z}$, then $F[(c+d\,x)^{1/n},\,x] = \frac{n}{d}\,\mathrm{Subst}[x^{n-1}\,F[x,\,-\frac{c}{d}+\frac{x^n}{d}],\,x,\,(c+d\,x)^{1/n}]\,\partial_x\,(c+d\,x)^{1/n}$

Rule: If $n\in\mathbb{Z}$, then

$$\int RF_x\,(a+b\,\mathrm{Log}[F[(c+d\,x)^{1/n},\,x]])\,dx \to \frac{n}{d}\,\mathrm{Subst}\left[\int x^{n-1}\,\mathrm{Subst}\left[RF_x,\,x,\,-\frac{c}{d}+\frac{x^n}{d}\right]\left(a+b\,F\left[x,\,-\frac{c}{d}+\frac{x^n}{d}\right]\right)dx,\,x,\,(c+d\,x)^{1/n}\right]$$

Program code:

```
Int[RFx_*(a_.+b_.*Log[u_]),x_Symbol] :=
  With[{lst=SubstForFractionalPowerOfLinear[RFx*(a+b*Log[u]),x]},
  lst[[2]]*lst[[4]]*Subst[Int[lst[[1]],x],x,lst[[3]]^(1/lst[[2]])] /;
 Not[FalseQ[lst]]] /;
FreeQ[{a,b},x] && RationalFunctionQ[RFx,x]
```

5. $\int (f+g\,x)^m \,\text{Log}\left[d+e\left(F^{c\,(a+b\,x)}\right)^n\right] dx$

1: $\int (f+g\,x)^m \,\text{Log}\left[1+e\left(F^{c\,(a+b\,x)}\right)^n\right] dx$ when $m>0$

Derivation: Integration by parts

Basis: $\text{Log}\left[1+e\left(F^{c\,(a+b\,x)}\right)^n\right] == -\partial_x \frac{\text{PolyLog}\left[2,-e\left(F^{c\,(a+b\,x)}\right)^n\right]}{b\,c\,n\,\text{Log}[F]}$

Rule: If $m>0$, then

$$\int (f+g\,x)^m \,\text{Log}\left[1+e\left(F^{c\,(a+b\,x)}\right)^n\right] dx \to -\frac{(f+g\,x)^m\,\text{PolyLog}\left[2,-e\left(F^{c\,(a+b\,x)}\right)^n\right]}{b\,c\,n\,\text{Log}[F]} + \frac{g\,m}{b\,c\,n\,\text{Log}[F]}\int (f+g\,x)^{m-1}\,\text{PolyLog}\left[2,-e\left(F^{c\,(a+b\,x)}\right)^n\right] dx$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*Log[1+e_.*(F_^(c_.*(a_.+b_.*x_)))^n_.],x_Symbol] :=
  -(f+g*x)^m*PolyLog[2,-e*(F^(c*(a+b*x)))^n]/(b*c*n*Log[F]) +
  g*m/(b*c*n*Log[F])*Int[(f+g*x)^(m-1)*PolyLog[2,-e*(F^(c*(a+b*x)))^n],x] /;
FreeQ[{F,a,b,c,e,f,g,n},x] && GtQ[m,0]
```

2: $\int (f+g\,x)^m \,\text{Log}\left[d+e\left(F^{c\,(a+b\,x)}\right)^n\right] dx$ when $m>0 \land d\neq 1$

Derivation: Integration by parts

Basis: $\partial_x \text{Log}[d+e\,g[x]] == \partial_x \text{Log}\left[1+\frac{e}{d}\,g[x]\right]$

Rule: If $m>0 \land d\neq 1$, then

$$\int (f+g\,x)^m \,\text{Log}\left[d+e\left(F^{c\,(a+b\,x)}\right)^n\right] dx \to \frac{(f+g\,x)^{m+1}\,\text{Log}\left[d+e\left(F^{c\,(a+b\,x)}\right)^n\right]}{g\,(m+1)} - \frac{(f+g\,x)^{m+1}\,\text{Log}\left[1+\frac{e}{d}\left(F^{c\,(a+b\,x)}\right)^n\right]}{g\,(m+1)} + \int (f+g\,x)^m\,\text{Log}\left[1+\frac{e}{d}\left(F^{c\,(a+b\,x)}\right)^n\right] dx$$

Program code:

```
Int[(f_.+g_.*x_)^m_.*Log[d_+e_.*(F_^(c_.*(a_.+b_.*x_)))^n_.],x_Symbol] :=
  (f+g*x)^(m+1)*Log[d+e*(F^(c*(a+b*x)))^n]/(g*(m+1)) -
  (f+g*x)^(m+1)*Log[1+e/d*(F^(c*(a+b*x)))^n]/(g*(m+1)) +
  Int[(f+g*x)^m*Log[1+e/d*(F^(c*(a+b*x)))^n],x] /;
FreeQ[{F,a,b,c,d,e,f,g,n},x] && GtQ[m,0] && NeQ[d,1]
```

6. $\int u\,\text{Log}\left[d+e\,x+f\sqrt{a+b\,x+c\,x^2}\right] dx$ when $e^2-c\,f^2 == 0$

1: $\int \text{Log}\left[d+e\,x+f\sqrt{a+b\,x+c\,x^2}\right] dx$ when $e^2-c\,f^2 == 0$

Derivation: Integration by parts and algebraic simplification

Rule: If $e^2-c\,f^2 == 0$, then $\frac{b\,f+2\,c\,f\,x+2\,e\sqrt{a+b\,x+c\,x^2}}{f\,(a+b\,x+c\,x^2)+(d+e\,x)\sqrt{a+b\,x+c\,x^2}} == -\frac{f^2\,(b^2-4\,a\,c)}{(2\,d\,e-b\,f^2)\,(a+b\,x+c\,x^2)-f\,(b\,d-2\,a\,e+(2\,c\,d-b\,e)\,x)\sqrt{a+b\,x+c\,x^2}}$.

Rule: If $e^2-c\,f^2 == 0$, then

$$\int \text{Log}\left[d+e\,x+f\sqrt{a+b\,x+c\,x^2}\right] dx \to x\,\text{Log}\left[d+e\,x+f\sqrt{a+b\,x+c\,x^2}\right] - \frac{1}{2}\int \frac{x\left(b\,f+2\,c\,f\,x+2\,e\sqrt{a+b\,x+c\,x^2}\right)}{f\,(a+b\,x+c\,x^2)+(d+e\,x)\sqrt{a+b\,x+c\,x^2}} dx$$

$$\longrightarrow x \operatorname{Log}\left[d+e x+f \sqrt{a+b x+c x^2}\right]+\frac{f^2\left(b^2-4 a c\right)}{2} \int\left(x \Big/ \left(\left(2 d e-b f^2\right)\left(a+b x+c x^2\right)-f\left(b d-2 a e+(2 c d-b e) x\right) \sqrt{a+b x+c x^2}\right)\right) dx$$

Program code:

```
Int[Log[d_.+e_.*x_+f_.*Sqrt[a_.+b_.*x_+c_.*x_^2]],x_Symbol] :=
  x*Log[d+e*x+f*Sqrt[a+b*x+c*x^2]] +
  f^2*(b^2-4*a*c)/2*Int[x/((2*d*e-b*f^2)*(a+b*x+c*x^2)-f*(b*d-2*a*e+(2*c*d-b*e)*x)*Sqrt[a+b*x+c*x^2]),x] /;
FreeQ[{a,b,c,d,e,f},x] && EqQ[e^2-c*f^2,0]
```

```
Int[Log[d_.+e_.*x_+f_.*Sqrt[a_.+c_.*x_^2]],x_Symbol] :=
  x*Log[d+e*x+f*Sqrt[a+c*x^2]] -
  a*c*f^2*Int[x/(d*e*(a+c*x^2)+f*(a*e-c*d*x)*Sqrt[a+c*x^2]),x] /;
FreeQ[{a,c,d,e,f},x] && EqQ[e^2-c*f^2,0]
```

2: $\int (g\,x)^m \operatorname{Log}\left[d+e\,x+f\sqrt{a+b\,x+c\,x^2}\right] dx$ when $e^2-c\,f^2==0 \wedge m\neq -1$

Derivation: Integration by parts and algebraic simplification

Rule: If $e^2-c\,f^2==0$, then $\frac{b\,f+2\,c\,f\,x+2\,e\,\sqrt{a+b\,x+c\,x^2}}{f\,(a+b\,x+c\,x^2)+(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}==-\frac{f^2\,(b^2-4\,a\,c)}{(2\,d\,e-b\,f^2)\,(a+b\,x+c\,x^2)-f\,(b\,d-2\,a\,e+(2\,c\,d-b\,e)\,x)\,\sqrt{a+b\,x+c\,x^2}}$.

Rule: If $e^2-c\,f^2==0 \wedge m\neq -1$, then

$$\int (g\,x)^m \operatorname{Log}\left[d+e\,x+f\sqrt{a+b\,x+c\,x^2}\right] dx \to \frac{(g\,x)^{m+1}\operatorname{Log}\left[d+e\,x+f\sqrt{a+b\,x+c\,x^2}\right]}{g\,(m+1)}-\frac{1}{2\,g\,(m+1)}\int\frac{(g\,x)^{m+1}\left(b\,f+2\,c\,f\,x+2\,e\,\sqrt{a+b\,x+c\,x^2}\right)}{f\,(a+b\,x+c\,x^2)+(d+e\,x)\,\sqrt{a+b\,x+c\,x^2}}\,dx$$

$$\to \frac{(g\,x)^{m+1}\operatorname{Log}\left[d+e\,x+f\sqrt{a+b\,x+c\,x^2}\right]}{g\,(m+1)}+\frac{f^2\,(b^2-4\,a\,c)}{2\,g\,(m+1)}\int\left((g\,x)^{m+1}\Big/\left((2\,d\,e-b\,f^2)\,(a+b\,x+c\,x^2)-f\,(b\,d-2\,a\,e+(2\,c\,d-b\,e)\,x)\,\sqrt{a+b\,x+c\,x^2}\right)\right)dx$$

Program code:

```
Int[(g_.*x_)^m_.*Log[d_.+e_.*x_+f_.*Sqrt[a_.+b_.*x_+c_.*x_^2]],x_Symbol] :=
  (g*x)^(m+1)*Log[d+e*x+f*Sqrt[a+b*x+c*x^2]]/(g*(m+1)) +
  f^2*(b^2-4*a*c)/(2*g*(m+1))*Int[(g*x)^(m+1)/((2*d*e-b*f^2)*(a+b*x+c*x^2)-f*(b*d-2*a*e+(2*c*d-b*e)*x)*Sqrt[a+b*x+c*x^2]),x] /;
FreeQ[{a,b,c,d,e,f,g,m},x] && EqQ[e^2-c*f^2,0] && NeQ[m,-1] && IntegerQ[2*m]
```

```
Int[(g_.*x_)^m_.*Log[d_.+e_.*x_+f_.*Sqrt[a_.+c_.*x_^2]],x_Symbol] :=
  (g*x)^(m+1)*Log[d+e*x+f*Sqrt[a+c*x^2]]/(g*(m+1)) -
  a*c*f^2/(g*(m+1))*Int[(g*x)^(m+1)/(d*e*(a+c*x^2)+f*(a*e-c*d*x)*Sqrt[a+c*x^2]),x] /;
FreeQ[{a,c,d,e,f,g,m},x] && EqQ[e^2-c*f^2,0] && NeQ[m,-1] && IntegerQ[2*m]
```

```
Int[v_.*Log[d_.+e_.*x_+f_.*Sqrt[u_]],x_Symbol] :=
  Int[v*Log[d+e*x+f*Sqrt[ExpandToSum[u,x]]],x] /;
FreeQ[{d,e,f},x] && QuadraticQ[u,x] && Not[QuadraticMatchQ[u,x]] && (EqQ[v,1] || MatchQ[v,(g_.*x)^m_. /; FreeQ[{g,m},x]])
```

7. $\int v \,\mathrm{Log}[u]\, dx$ when u is free of inverse functions

1: $\int \mathrm{Log}[u]\, dx$ when u is free of inverse functions

Reference: A&S 4.1.53

Derivation: Integration by parts

Rule: If InverseFunctionFreeQ[u, x], then

$$\int \mathrm{Log}[u]\, dx \longrightarrow x\,\mathrm{Log}[u] - \int \frac{x\, \partial_x u}{u}\, dx$$

Program code:

```
Int[Log[u_],x_Symbol] :=
  x*Log[u] - Int[SimplifyIntegrand[x*D[u,x]/u,x],x] /;
InverseFunctionFreeQ[u,x]
```

```
Int[Log[u_],x_Symbol] :=
  x*Log[u] - Int[SimplifyIntegrand[x*Simplify[D[u,x]/u],x],x] /;
ProductQ[u]
```

2. $\int (a+b\,x)^m \,\text{Log}[u]\, dx$ when u is free of inverse functions

1: $\int \frac{\text{Log}[u]}{a+b\,x}\, dx$ when $\text{RationalFunctionQ}\left[\frac{\partial_x u}{u}, x\right]$

Reference: G&R 2.727.2

Derivation: Integration by parts

Basis: $\frac{1}{a+b\,x} == \partial_x \frac{\text{Log}[a+b\,x]}{b}$

Rule: If $\text{RationalFunctionQ}\left[\frac{\partial_x u}{u}, x\right]$, then

$$\int \frac{\text{Log}[u]}{a+b\,x}\, dx \longrightarrow \frac{\text{Log}[a+b\,x]\,\text{Log}[u]}{b} - \frac{1}{b}\int \frac{\text{Log}[a+b\,x]\,\partial_x u}{u}\, dx$$

Program code:

```
Int[Log[u_]/(a_.+b_.*x_),x_Symbol] :=
  Log[a+b*x]*Log[u]/b -
  1/b*Int[SimplifyIntegrand[Log[a+b*x]*D[u,x]/u,x],x] /;
FreeQ[{a,b},x] && RationalFunctionQ[D[u,x]/u,x] && (NeQ[a,0] || Not[BinomialQ[u,x] && EqQ[BinomialDegree[u,x]^2,1]])
```

2: $\int (a+b\,x)^m \,\text{Log}[u]\, dx$ when $\text{InverseFunctionFreeQ}[u, x] \land m \neq -1$

Reference: G&R 2.725.1, A&S 4.1.54

Derivation: Integration by parts

Basis: $(a+b\,x)^m == \partial_x \frac{(a+b\,x)^{m+1}}{b\,(m+1)}$

Rule: If $\text{InverseFunctionFreeQ}[u, x] \land m \neq -1$, then

$$\int (a+b\,x)^m \,\text{Log}[u] \, dx \longrightarrow \frac{(a+b\,x)^{m+1}\,\text{Log}[u]}{b\,(m+1)} - \frac{1}{b\,(m+1)} \int \frac{(a+b\,x)^{m+1}\,\partial_x u}{u} \, dx$$

Program code:

```
Int[(a_.+b_.*x_)^m_.*Log[u_],x_Symbol] :=
  (a+b*x)^(m+1)*Log[u]/(b*(m+1)) -
  1/(b*(m+1))*Int[SimplifyIntegrand[(a+b*x)^(m+1)*D[u,x]/u,x],x] /;
FreeQ[{a,b,m},x] && InverseFunctionFreeQ[u,x] && NeQ[m,-1] (* && Not[FunctionOfQ[x^(m+1),u,x]] && FalseQ[PowerVariableExpn[u,m+1,x]] *
```

3: $\int \frac{\text{Log}[u]}{Q_x} \, dx$ when $\text{QuadraticQ}[Q_x] \land \text{InverseFunctionFreeQ}[u,\ x]$

Derivation: Integration by parts

Rule: If $\text{QuadraticQ}[Q_x] \land \text{InverseFunctionFreeQ}[u,\ x]$, let $v = \int \frac{1}{Q_x} \, dx$, then

$$\int \frac{\text{Log}[u]}{Q_x} \, dx \longrightarrow v\,\text{Log}[u] - \int \frac{v\,\partial_x u}{u} \, dx$$

Program code:

```
Int[Log[u_]/Qx_,x_Symbol] :=
  With[{v=IntHide[1/Qx,x]},
  v*Log[u] - Int[SimplifyIntegrand[v*D[u,x]/u,x],x]] /;
QuadraticQ[Qx,x] && InverseFunctionFreeQ[u,x]
```

4: $\int u^{a x} \operatorname{Log}[u] \, dx$ when u is free of inverse functions

Basis: $u^{a x} \operatorname{Log}[u] == \frac{\partial_x u^{a x}}{a} - x\, u^{a x-1}\, \partial_x u$

Rule: If InverseFunctionFreeQ[u, x], then

$$\int u^{a x} \operatorname{Log}[u] \, dx \longrightarrow \frac{u^{a x}}{a} - \int x\, u^{a x-1}\, \partial_x u \, dx$$

Program code:

```
Int[u_^(a_.*x_)*Log[u_],x_Symbol] :=
  u^(a*x)/a - Int[SimplifyIntegrand[x*u^(a*x-1)*D[u,x],x],x] /;
FreeQ[a,x] && InverseFunctionFreeQ[u,x]
```

5: $\int v \operatorname{Log}[u] \, dx$ when u and $\int v \, dx$ are free of inverse functions

Derivation: Integration by parts

Rule: If InverseFunctionFreeQ[u, x], let $w = \int v \, dx$, if InverseFunctionFreeQ[w, x], then

$$\int v \operatorname{Log}[u] \, dx \longrightarrow w \operatorname{Log}[u] - \frac{1}{b} \int \frac{w\, \partial_x u}{u} \, dx$$

Program code:

```
Int[v_*Log[u_],x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[Log[u],w,x] - Int[SimplifyIntegrand[w*D[u,x]/u,x],x] /;
 InverseFunctionFreeQ[w,x]] /;
InverseFunctionFreeQ[u,x]
```

```
Int[v_*Log[u_],x_Symbol] :=
  With[{w=IntHide[v,x]},
  Dist[Log[u],w,x] - Int[SimplifyIntegrand[w*Simplify[D[u,x]/u],x],x] /;
 InverseFunctionFreeQ[w,x]] /;
ProductQ[u]
```

8. $\int u \,\mathrm{Log}[v]\, \mathrm{Log}[w]\, dx$ when v, w and $\int u\, dx$ are free of inverse functions

1: $\int \mathrm{Log}[v]\, \mathrm{Log}[w]\, dx$ when v and w are free of inverse functions

Derivation: Integration by parts

Rule: If v and w are free of inverse functions, then

$$\int \mathrm{Log}[v]\, \mathrm{Log}[w]\, dx \longrightarrow x\, \mathrm{Log}[v]\, \mathrm{Log}[w] - \int \frac{x\, \mathrm{Log}[w]\, \partial_x v}{v}\, dx - \int \frac{x\, \mathrm{Log}[v]\, \partial_x w}{w}\, dx$$

Program code:

```
Int[Log[v_]*Log[w_],x_Symbol] :=
  x*Log[v]*Log[w] -
  Int[SimplifyIntegrand[x*Log[w]*D[v,x]/v,x],x] -
  Int[SimplifyIntegrand[x*Log[v]*D[w,x]/w,x],x] /;
InverseFunctionFreeQ[v,x] && InverseFunctionFreeQ[w,x]
```

2: $\int u \,\text{Log}[v]\, \text{Log}[w]\, dx$ when v, w and $\int u\, dx$ are free of inverse functions

Derivation: Integration by parts

Rule: If v and w are free of inverse functions, let $z == \int u\, dx$, if z is free of inverse functions, then

$$\int u \,\text{Log}[v]\, \text{Log}[w]\, dx \longrightarrow z \,\text{Log}[v]\, \text{Log}[w] - \int \frac{z \,\text{Log}[w]\, \partial_x v}{v}\, dx - \int \frac{z \,\text{Log}[v]\, \partial_x w}{w}\, dx$$

Program code:

```
Int[u_*Log[v_]*Log[w_],x_Symbol] :=
  With[{z=IntHide[u,x]},
  Dist[Log[v]*Log[w],z,x] -
  Int[SimplifyIntegrand[z*Log[w]*D[v,x]/v,x],x] -
  Int[SimplifyIntegrand[z*Log[v]*D[w,x]/w,x],x] /;
 InverseFunctionFreeQ[z,x]] /;
InverseFunctionFreeQ[v,x] && InverseFunctionFreeQ[w,x]
```

9: $\int f^{a \operatorname{Log}[u]} \, dx$

Derivation: Algebraic simplification

Basis: $f^{a \operatorname{Log}[g]} = g^{a \operatorname{Log}[f]}$

Rule:

$$\int f^{a \operatorname{Log}[u]} \, dx \longrightarrow \int u^{a \operatorname{Log}[f]} \, dx$$

Program code:

```
Int[f_^(a_.*Log[u_]),x_Symbol] :=
  Int[u^(a*Log[f]),x] /;
FreeQ[{a,f},x]
```

10: $\int \frac{F[\text{Log}[a\,x^n]]}{x}\,dx$

Derivation: Integration by substitution

Basis: $\frac{F[\text{Log}[a\,x^n]]}{x} = \frac{1}{n}\,F[\text{Log}[a\,x^n]]\,\partial_x\,\text{Log}[a\,x^n]$

Rule:

$$\int \frac{F[\text{Log}[a\,x^n]]}{x}\,dx \longrightarrow \frac{1}{n}\,\text{Subst}\left[\int F[x]\,dx,\,x,\,\text{Log}[a\,x^n]\right]$$

Program code:

```
(* If[TrueQ[$LoadShowSteps],

Int[u_/x_,x_Symbol] :=
  With[{lst=FunctionOfLog[u,x]},
  ShowStep["","Int[F[Log[a*x^n]]/x,x]","Subst[Int[F[x],x],x,Log[a*x^n]]/n",Hold[
  1/lst[[3]]*Subst[Int[lst[[1]],x],x,Log[lst[[2]]]]]] /;
 Not[FalseQ[lst]]] /;
SimplifyFlag && NonsumQ[u],

Int[u_/x_,x_Symbol] :=
  With[{lst=FunctionOfLog[u,x]},
  1/lst[[3]]*Subst[Int[lst[[1]],x],x,Log[lst[[2]]]] /;
 Not[FalseQ[lst]]] /;
NonsumQ[u]] *)
```

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  With[{lst=FunctionOfLog[Cancel[x*u],x]},
  ShowStep["","Int[F[Log[a*x^n]]/x,x]","Subst[Int[F[x],x],x,Log[a*x^n]]/n",Hold[
  1/lst[[3]]*Subst[Int[lst[[1]],x],x,Log[lst[[2]]]]]] /;
 Not[FalseQ[lst]]] /;
SimplifyFlag && NonsumQ[u],

Int[u_,x_Symbol] :=
  With[{lst=FunctionOfLog[Cancel[x*u],x]},
  1/lst[[3]]*Subst[Int[lst[[1]],x],x,Log[lst[[2]]]] /;
 Not[FalseQ[lst]]] /;
NonsumQ[u]]
```

11: $\int u\,\text{Log}[\text{Gamma}[v]]\,dx$

Derivation: Piecewise constant extraction

Basis: $\partial_x\,(\text{Log}[\text{Gamma}[F[x]]] - \text{LogGamma}[F[x]]) == 0$

Rule:

$$\int u\,\text{Log}[\text{Gamma}[v]]\,dx \longrightarrow (\text{Log}[\text{Gamma}[v]] - \text{LogGamma}[v])\int u\,dx + \int u\,\text{LogGamma}[v]\,dx$$

Program code:

```
Int[u_.*Log[Gamma[v_]],x_Symbol] :=
  (Log[Gamma[v]]-LogGamma[v])*Int[u,x] + Int[u*LogGamma[v],x]
```

N: $\int u\left(a\,w + b\,w\,\text{Log}[v]^n\right)^p\,dx$ when $p \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}$, then

$$\int u\left(a\,w + b\,w\,\text{Log}[v]^n\right)^p\,dx \longrightarrow \int u\,w^p\left(a + b\,\text{Log}[v]^n\right)^p\,dx$$

Program code:

```
Int[u_.*(a_.*w_+b_.*w_*Log[v_]^n_.)^p_.,x_Symbol] :=
  Int[u*w^p*(a+b*Log[v]^n)^p,x] /;
FreeQ[{a,b,n},x] && IntegerQ[p]
```