Integrand Simplification Rules

1. $\int u\,(v+w)^p\,dx$ when $v == 0$

x: $\int u\,(v+w)^p\,dx$ when $v == 0$

Derivation: Algebraic simplification

Note: Many rules assume coefficients are not unrecognized zeros.

Note: Unfortunately this rule is commented out because it is too inefficient.

Rule: If $v == 0$, then

$$\int u\,(v+w)^p\,dx \longrightarrow \int u\,w^p\,dx$$

Program code:

```
(* Int[u_.*(v_+w_)^p_.,x_Symbol] :=
  Int[u*w^p,x] /;
FreeQ[p,x] && EqQ[v,0] *)
```

1: $\int u\,(a+b\,x^n)^p\,dx$ when $a == 0$

Derivation: Algebraic simplification

Rule: If $a == 0$, then

$$\int u\,(a+b\,x^n)^p\,dx \longrightarrow \int u\,(b\,x^n)^p\,dx$$

Program code:

```
Int[u_.*(a_+b_.*x_^n_.)^p_.,x_Symbol] :=
  Int[u*(b*x^n)^p,x] /;
FreeQ[{a,b,n,p},x] && EqQ[a,0]
```

2: $\int u\,(a+b\,x^n)^p\,dx$ when $b == 0$

Derivation: Algebraic simplification

Rule: If $b == 0$, then

$$\int u\,(a+b\,x^n)^p\,dx \longrightarrow \int u\,a^p\,dx$$

Program code:

```
Int[u_.*(a_.+b_.*x_^n_.)^p_.,x_Symbol] :=
  Int[u*a^p,x] /;
FreeQ[{a,b,n,p},x] && EqQ[b,0]
```

3: $\int u\,(a+b\,x^n+c\,x^{2\,n})^p\,dx$ when $a == 0$

Derivation: Algebraic simplification

Rule: If $a == 0$, then

$$\int u\,(a+b\,x^n+c\,x^{2\,n})^p\,dx \longrightarrow \int u\,(b\,x^n+c\,x^{2\,n})^p\,dx$$

Program code:

```
Int[u_.*(a_+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  Int[u*(b*x^n+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[j,2*n] && EqQ[a,0]
```

4: $\int u\,(a+b\,x^n+c\,x^{2\,n})^p\,dx$ when $b == 0$

Derivation: Algebraic simplification

Rule: If $b == 0$, then

$$\int u\,(a+b\,x^n+c\,x^{2\,n})^p\,dx \longrightarrow \int u\,(a+c\,x^{2\,n})^p\,dx$$

Program code:

```
Int[u_.*(a_.+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  Int[u*(a+c*x^(2*n))^p,x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[j,2*n] && EqQ[b,0]
```

5: $\int u\,(a+b\,x^n+c\,x^{2\,n})^p\,dx$ when $c == 0$

Derivation: Algebraic simplification

Rule: If $c == 0$, then

$$\int u\,(a+b\,x^n+c\,x^{2\,n})^p\,dx \longrightarrow \int u\,(a+b\,x^n)^p\,dx$$

Program code:

```
Int[u_.*(a_.+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  Int[u*(a+b*x^n)^p,x] /;
FreeQ[{a,b,c,n,p},x] && EqQ[j,2*n] && EqQ[c,0]
```

2: $\int u\,(a\,v+b\,v+w)^p\,dx$ when v depends on x

Derivation: Algebraic simplification

Rule: If v depends on x, then

$$\int u\,(a\,v+b\,v+w)^p\,dx \longrightarrow \int u\,((a+b)\,v+w)^p\,dx$$

Program code:

```
Int[u_.*(a_.*v_+b_.*v_+w_.)^p_.,x_Symbol] :=
  Int[u*((a+b)*v+w)^p,x] /;
FreeQ[{a,b},x] && Not[FreeQ[v,x]]
```

3: $\int u\, P_m[x]^p \, dx$ when $p \notin \mathbb{Q} \land \text{Simplify}[p] \in \mathbb{Q}$

Derivation: Algebraic simplification

Note: Integration rules assume integer and rational exponents are recognized as such.

Rule: If $p \notin \mathbb{Q} \land \text{Simplify}[p] \in \mathbb{Q}$, then

$$\int u\, P_m[x]^p \, u \, dx \longrightarrow \int u\, P_m[x]^{\text{Simplify}[p]} \, dx$$

Program code:

```
Int[u_.*Pm_^p_,x_Symbol] :=
  Int[u*Pm^Simplify[p],x] /;
PolyQ[Pm,x] && Not[RationalQ[p]] && FreeQ[p,x] && RationalQ[Simplify[p]]
```

4. $\int a\, u \, dx$

1: $\int a \, dx$

Reference: CRC 1

Rule:

$$\int a \, dx \longrightarrow a\, x$$

Program code:

```
Int[a_,x_Symbol] :=
   a*x /;
FreeQ[a,x]
```

2: $\int a\,(b+c\,x)\,dx$

Derivation: Power rule for integration

Rule:

$$\int a\,(b+c\,x)\,dx \longrightarrow \frac{a\,(b+c\,x)^2}{2\,c}$$

Program code:

```
Int[a_*(b_+c_.*x_),x_Symbol] :=
  a*(b+c*x)^2/(2*c) /;
FreeQ[{a,b,c},x]
```

3: $\int a\,u\,dx$

Reference: G&R 2.02.1, CRC 2

Derivation: Constant extraction

Note: Since the rule for extracting the imaginary unit from integrands includes the function `Identity`, it is not displayed when showing steps thus avoiding trivial steps when integrating expressions involving hyperbolic functions.

Rule:

$$\int a\,u\,dx \longrightarrow a\int u\,dx$$

Program code:

```
Int[-u_,x_Symbol] :=
  Identity[-1]*Int[u,x]
```

```
Int[Complex[0,a_]*u_,x_Symbol] :=
  Complex[Identity[0],a]*Int[u,x] /;
FreeQ[a,x] && EqQ[a^2,1]
```

```
Int[a_*u_,x_Symbol] :=
  a*Int[u,x] /;
FreeQ[a,x] && Not[MatchQ[u, b_*v_ /; FreeQ[b,x]]]
```

5: $\int a\,u + b\,v + \cdots \, dx$

Reference: G&R 2.02.2, 2.111.1 CRC 2, 4, 23, 27

Note: By actually integrating linear power of x terms, this rule eliminates numerous trivial integration steps.

Rule:

$$\int a\,u + b\,v + \cdots \, dx \longrightarrow a \int u \, dx + b \int v \, dx + \cdots$$

Program code:

```
If[TrueQ[$LoadShowSteps],

Int[u_,x_Symbol] :=
  ShowStep["","Int[a*u + b*v + …,x]","a*Integrate[u,x] + b*Integrate[v,x] + …",Hold[
  IntSum[u,x]]] /;
SimplifyFlag && SumQ[u],

Int[u_,x_Symbol] :=
  IntSum[u,x] /;
SumQ[u]]
```

6: $\int (c\,x)^m\,(u+v+\cdots)\,dx$

Derivation: Algebraic expansion

Rule:

$$\int (c\,x)^m\,(u+v+\cdots)\,dx \longrightarrow \int (c\,x)^m\,u+(c\,x)^m\,v+\cdots\,dx$$

Program code:

```
Int[(c_.*x_)^m_.*u_,x_Symbol] :=
  Int[ExpandIntegrand[(c*x)^m*u,x],x] /;
FreeQ[{c,m},x] && SumQ[u] && Not[LinearQ[u,x]] && Not[MatchQ[u,a_+b_.*v_ /; FreeQ[{a,b},x] && InverseFunctionQ[v]]]
```

?: $\int u\,(a\,x^n)^m\,dx$ when $m \notin \mathbb{Z} \land m\,n \in \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(a\,x^n)^m}{x^{m\,n}} == 0$

Rule: If $m \notin \mathbb{Z} \land m\,n \in \mathbb{Z}$, then

$$\int u\,(a\,x^n)^m\,x \longrightarrow \frac{a^{\text{IntPart}[m]}\,(a\,x^n)^{\text{FracPart}[m]}}{x^{n\,\text{FracPart}[m]}}\int u\,x^{m\,n}\,dx$$

Program code:

```
Int[u_.*(a_.*x_^n_)^m_,x_Symbol] :=
  a^IntPart[m]*(a*x^n)^FracPart[m]/x^(n*FracPart[m])*Int[u*x^(m*n),x] /;
FreeQ[{a,m,n},x] && Not[IntegerQ[m]]
```

7. $\int u\,(a\,v)^m\,(b\,v)^n\,dx$

1: $\int u\,v^m\,(b\,v)^n\,dx$ when $m\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $m\in\mathbb{Z}$, then $v^m == \frac{1}{b^m}\,(b\,v)^m$

Rule: If $m\in\mathbb{Z}$, then

$$\int u\,v^m\,(b\,v)^n\,dx \longrightarrow \frac{1}{b^m}\int u\,(b\,v)^{m+n}\,dx$$

Program code:

```
Int[u_.*v_^m_.*(b_*v_)^n_,x_Symbol] :=
  1/b^m*Int[u*(b*v)^(m+n),x] /;
FreeQ[{b,n},x] && IntegerQ[m]
```

2. $\int u\,(a\,v)^m\,(b\,v)^n\,dx$ when $m\notin\mathbb{Z}\wedge n\notin\mathbb{Z}$

1. $\int u\,(a\,v)^m\,(b\,v)^n\,dx$ when $m\notin\mathbb{Z}\wedge n+\frac{1}{2}\in\mathbb{Z}$

1. $\int u\,(a\,v)^m\,(b\,v)^n\,dx$ when $m\notin\mathbb{Z}\wedge n+\frac{1}{2}\in\mathbb{Z}^+$

1: $\int u\,(a\,v)^m\,(b\,v)^n\,dx$ when $m\notin\mathbb{Z}\wedge n+\frac{1}{2}\in\mathbb{Z}^+\wedge m+n\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{b\,F[x]}}{\sqrt{a\,F[x]}} == 0$

Basis: If $n+\frac{1}{2}\in\mathbb{Z}$, then $(b\,v)^n == \frac{b^{n-\frac{1}{2}}\sqrt{b\,v}}{a^{n-\frac{1}{2}}\sqrt{a\,v}}\,(a\,v)^n$

Rule: If $m\notin\mathbb{Z}\land n+\frac{1}{2}\in\mathbb{Z}^+\land m+n\in\mathbb{Z}$, then

$$\int u\,(a\,v)^m\,(b\,v)^n\,dx \longrightarrow \frac{a^{m+\frac{1}{2}}\,b^{n-\frac{1}{2}}\sqrt{b\,v}}{\sqrt{a\,v}}\int u\,v^{m+n}\,dx$$

Program code:

```
Int[u_.*(a_.*v_)^m_*(b_.*v_)^n_,x_Symbol] :=
  a^(m+1/2)*b^(n-1/2)*Sqrt[b*v]/Sqrt[a*v]*Int[u*v^(m+n),x] /;
FreeQ[{a,b,m},x] && Not[IntegerQ[m]] && IGtQ[n+1/2,0] && IntegerQ[m+n]
```

x: $\int u\,(a\,v)^m\,(b\,v)^n\,dx$ when $m\notin\mathbb{Z}\land n+\frac{1}{2}\in\mathbb{Z}^+\land m+n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{\sqrt{b\,F[x]}}{\sqrt{a\,F[x]}} == 0$

Basis: If $n+\frac{1}{2}\in\mathbb{Z}$, then $(b\,v)^n == \frac{b^{n-\frac{1}{2}}\sqrt{b\,v}}{a^{n-\frac{1}{2}}\sqrt{a\,v}}\,(a\,v)^n$

Rule: If $m\notin\mathbb{Z}\land n+\frac{1}{2}\in\mathbb{Z}^+\land m+n\notin\mathbb{Z}$, then

$$\int u\,(a\,v)^m\,(b\,v)^n\,dx \longrightarrow \frac{b^{n-\frac{1}{2}}\sqrt{b\,v}}{a^{n-\frac{1}{2}}\sqrt{a\,v}}\int u\,(a\,v)^{m+n}\,dx$$

Program code:

```
(* Int[u_.*(a_.*v_)^m_*(b_.*v_)^n_,x_Symbol] :=
  b^(n-1/2)*Sqrt[b*v]/(a^(n-1/2)*Sqrt[a*v])*Int[u*(a*v)^(m+n),x] /;
FreeQ[{a,b,m},x] && Not[IntegerQ[m]] && IGtQ[n+1/2,0] && Not[IntegerQ[m+n]] *)
```

2. $\int u\,(a\,v)^m\,(b\,v)^n\,dx$ when $m\notin\mathbb{Z}\land n-\frac{1}{2}\in\mathbb{Z}^-$

1: $\int u\,(a\,v)^m\,(b\,v)^n\,dx$ when $m\notin\mathbb{Z}\land n-\frac{1}{2}\in\mathbb{Z}^-\land m+n\in\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{\sqrt{a\,F[x]}}{\sqrt{b\,F[x]}}=0$

Basis: If $n-\frac{1}{2}\in\mathbb{Z}$, then $(b\,v)^n=\frac{b^{n+\frac{1}{2}}\sqrt{a\,v}}{a^{n+\frac{1}{2}}\sqrt{b\,v}}\,(a\,v)^n$

Rule: If $m\notin\mathbb{Z}\land n-\frac{1}{2}\in\mathbb{Z}^-\land m+n\in\mathbb{Z}$, then

$$\int u\,(a\,v)^m\,(b\,v)^n\,dx\to\frac{a^{m-\frac{1}{2}}\,b^{n+\frac{1}{2}}\sqrt{a\,v}}{\sqrt{b\,v}}\int u\,v^{m+n}\,dx$$

Program code:

```
Int[u_.*(a_.*v_)^m_*(b_.*v_)^n_,x_Symbol] :=
  a^(m-1/2)*b^(n+1/2)*Sqrt[a*v]/Sqrt[b*v]*Int[u*v^(m+n),x] /;
FreeQ[{a,b,m},x] && Not[IntegerQ[m]] && ILtQ[n-1/2,0] && IntegerQ[m+n]
```

x: $\int u\,(a\,v)^m\,(b\,v)^n\,dx$ when $m\notin\mathbb{Z}\land n-\frac{1}{2}\in\mathbb{Z}^-\land m+n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\frac{\sqrt{a\,F[x]}}{\sqrt{b\,F[x]}}==0$

Basis: If $n-\frac{1}{2}\in\mathbb{Z}$, then $(b\,v)^n==\frac{b^{n+\frac{1}{2}}\sqrt{a\,v}}{a^{n+\frac{1}{2}}\sqrt{b\,v}}(a\,v)^n$

Rule: If $m\notin\mathbb{Z}\land n-\frac{1}{2}\in\mathbb{Z}^-\land m+n\notin\mathbb{Z}$, then

$$\int u\,(a\,v)^m\,(b\,v)^n\,dx\longrightarrow\frac{b^{n+\frac{1}{2}}\sqrt{a\,v}}{a^{n+\frac{1}{2}}\sqrt{b\,v}}\int u\,(a\,v)^{m+n}\,dx$$

Program code:

```
(* Int[u_.*(a_.*v_)^m_*(b_.*v_)^n_,x_Symbol] :=
  b^(n+1/2)*Sqrt[a*v]/(a^(n+1/2)*Sqrt[b*v])*Int[u*(a*v)^(m+n),x] /;
FreeQ[{a,b,m},x] && Not[IntegerQ[m]] && ILtQ[n-1/2,0] && Not[IntegerQ[m+n]] *)
```

2. $\int u\,(a\,v)^m\,(b\,v)^n\,dx$ when $m \notin \mathbb{Z}\,\land\,n \notin \mathbb{Z}$

1: $\int u\,(a\,v)^m\,(b\,v)^n\,dx$ when $m \notin \mathbb{Z}\,\land\,n \notin \mathbb{Z}\,\land\,m+n \in \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(b\,F[x])^n}{(a\,F[x])^n} == 0$

Rule: If $m \notin \mathbb{Z}\,\land\,n \notin \mathbb{Z}\,\land\,m+n \in \mathbb{Z}$, then

$$\int u\,(a\,v)^m\,(b\,v)^n\,dx \longrightarrow \frac{a^{m+n}\,(b\,v)^n}{(a\,v)^n}\int u\,v^{m+n}\,dx$$

Program code:

```
Int[u_.*(a_.*v_)^m_*(b_.*v_)^n_,x_Symbol] :=
  a^(m+n)*(b*v)^n/(a*v)^n*Int[u*v^(m+n),x] /;
FreeQ[{a,b,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && IntegerQ[m+n]
```

2: $\int u\,(a\,v)^m\,(b\,v)^n\,dx$ when $m\notin\mathbb{Z}\land n\notin\mathbb{Z}\land m+n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(b\,F[x])^n}{(a\,F[x])^n} == 0$

Rule: If $m\notin\mathbb{Z}\land n\notin\mathbb{Z}\land m+n\notin\mathbb{Z}$, then

$$\int u\,(a\,v)^m\,(b\,v)^n\,dx \longrightarrow \frac{b^{\text{IntPart}[n]}\,(b\,v)^{\text{FracPart}[n]}}{a^{\text{IntPart}[n]}\,(a\,v)^{\text{FracPart}[n]}}\int u\,(a\,v)^{m+n}\,dx$$

Program code:

```
Int[u_.*(a_.*v_)^m_*(b_.*v_)^n_,x_Symbol] :=
  b^IntPart[n]*(b*v)^FracPart[n]/(a^IntPart[n]*(a*v)^FracPart[n])*Int[u*(a*v)^(m+n),x] /;
FreeQ[{a,b,m,n},x] && Not[IntegerQ[m]] && Not[IntegerQ[n]] && Not[IntegerQ[m+n]]
```

8. $\int u\,(a+b\,v)^m\,(c+d\,v)^n\,dx$ when $b\,c-a\,d==0$

1: $\int u\,(a+b\,v)^m\,(c+d\,v)^n\,dx$ when $b\,c-a\,d==0\ \land\ \left(m\in\mathbb{Z}\ \lor\ \frac{b}{d}>0\right)$

Derivation: Algebraic simplification

Basis: If $b\,c-a\,d==0\ \land\ \left(m\in\mathbb{Z}\ \lor\ \frac{b}{d}>0\right)$, then $(a+b\,z)^m==\left(\frac{b}{d}\right)^m(c+d\,z)^m$

Rule: If $b\,c-a\,d==0\ \land\ \left(m\in\mathbb{Z}\ \lor\ \frac{b}{d}>0\right)$, then

$$\int u\,(a+b\,v)^m\,(c+d\,v)^n\,dx\ \longrightarrow\ \left(\frac{b}{d}\right)^m\int u\,(c+d\,v)^{m+n}\,dx$$

Program code:

```
Int[u_.*(a_+b_.*v_)^m_.*(c_+d_.*v_)^n_.,x_Symbol] :=
  (b/d)^m*Int[u*(c+d*v)^(m+n),x] /;
FreeQ[{a,b,c,d,n},x] && EqQ[b*c-a*d,0] && IntegerQ[m] && (Not[IntegerQ[n]] || SimplerQ[c+d*x,a+b*x])
```

```
Int[u_.*(a_+b_.*v_)^m_*(c_+d_.*v_)^n_,x_Symbol] :=
  (b/d)^m*Int[u*(c+d*v)^(m+n),x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[b*c-a*d,0] && GtQ[b/d,0] && Not[IntegerQ[m] || IntegerQ[n]]
```

2: $\int u\,(a+b\,v)^m\,(c+d\,v)^n\,dx$ when $b\,c-a\,d==0\ \land\ \neg\left(m\in\mathbb{Z}\ \lor\ n\in\mathbb{Z}\ \lor\ \frac{b}{d}>0\right)$

Derivation: Piecewise constant extraction

Basis: If $b\,c-a\,d==0$, then $\partial_x\frac{(a+b\,F[x])^m}{(c+d\,F[x])^m}==0$

Rule: If $b\,c-a\,d==0\ \land\ \neg\left(m\in\mathbb{Z}\ \lor\ n\in\mathbb{Z}\ \lor\ \frac{b}{d}>0\right)$, then

$$\int u\,(a+b\,v)^m\,(c+d\,v)^n\,dx \longrightarrow \frac{(a+b\,v)^m}{(c+d\,v)^m}\int u\,(c+d\,v)^{m+n}\,dx$$

Program code:

```
Int[u_.*(a_+b_.*v_)^m_*(c_+d_.*v_)^n_,x_Symbol] :=
  (a+b*v)^m/(c+d*v)^m*Int[u*(c+d*v)^(m+n),x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[b*c-a*d,0] && Not[IntegerQ[m] || IntegerQ[n] || GtQ[b/d,0]]
```

x. $\int u\,(a+b\,v)^m\,(c+d\,v)^m\,dx$ when $b\,c+a\,d=0$

1: $\int u\,(a+b\,v)^m\,(c+d\,v)^m\,dx$ when $b\,c+a\,d=0 \land (m\in\mathbb{Z} \lor a>0 \land c>0)$

Derivation: Algebraic simplification

Basis: If $b\,c+a\,d=0 \land (m\in\mathbb{Z} \lor a>0 \land c>0)$, then $(a+b\,v)^m\,(c+d\,v)^m = (a\,c+b\,d\,v^2)^m$

Rule: If $b\,c+a\,d=0 \land (m\in\mathbb{Z} \lor a>0 \land c>0)$, then

$$\int u\,(a+b\,v)^m\,(c+d\,v)^m\,dx \longrightarrow \int u\,(a\,c+b\,d\,v^2)^m\,dx$$

Program code:

```
(* Int[u_.*(a_+b_.*v_)^m_.*(c_+d_.*v_)^m_.,x_Symbol] :=
  Int[u*(a*c+b*d*v^2)^m,x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[b*c+a*d,0] && (IntegerQ[m] || GtQ[a,0] && GtQ[c,0]) &&
  (Not[AlgebraicFunctionQ[u,x]] || Not[MatchQ[v,e_.*x^n_. /; FreeQ[{e,n},x]]]) *)
```

2: $\int u\,(a+b\,v)^m\,(c+d\,v)^m\,dx$ when $b\,c+a\,d==0\land m\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: If $b\,c+a\,d==0$, then $\partial_x\frac{(a+b\,F[x])^m\,(c+d\,F[x])^m}{(a\,c+b\,d\,F[x]^2)^m}==0$

Rule: If $b\,c+a\,d==0\land m\notin\mathbb{Z}$, then

$$\int u\,(a+b\,v)^m\,(c+d\,v)^m\,dx\,\longrightarrow\,\frac{(a+b\,v)^{\text{FracPart}[m]}\,(c+d\,v)^{\text{FracPart}[m]}}{(a\,c+b\,d\,v^2)^{\text{FracPart}[m]}}\int u\,(a\,c+b\,d\,v^2)^m\,dx$$

Program code:

```
(* Int[u_.*(a_+b_.*v_)^m_*(c_+d_.*v_)^m_,x_Symbol] :=
  (a+b*v)^FracPart[m]*(c+d*v)^FracPart[m]/(a*c+b*d*v^2)^FracPart[m]*Int[u*(a*c+b*d*v^2)^m,x] /;
FreeQ[{a,b,c,d,m},x] && EqQ[b*c+a*d,0] && Not[IntegerQ[m]] &&
  (Not[AlgebraicFunctionQ[u,x]] || Not[MatchQ[v,e_.*x^n_. /; FreeQ[{e,n},x]]]) *)
```

9: $\int u\,(a+b\,v)^m\,(A+B\,v+C\,v^2)\,dx$ when $A\,b^2-a\,b\,B+a^2\,C=0\,\wedge\,m\le -1$

Derivation: Algebraic simplification

Basis: If $A\,b^2-a\,b\,B+a^2\,C=0$, then $A+B\,z+C\,z^2=\frac{1}{b^2}\,(a+b\,z)\,(b\,B-a\,C+b\,C\,z)$

Rule: If $A\,b^2-a\,b\,B+a^2\,C=0\,\wedge\,m\le -1$, then

$$\int u\,(a\,v)^m\,(b\,v+c\,v^2)\,dx\,\longrightarrow\,\frac{1}{a}\int u\,(a\,v)^{m+1}\,(b+c\,v)\,dx$$

$$\int u\,(a+b\,v)^m\,(A+B\,v+C\,v^2)\,dx\,\longrightarrow\,\frac{1}{b^2}\int u\,(a+b\,v)^{m+1}\,(b\,B-a\,C+b\,C\,v)\,dx$$

Program code:

```
Int[u_.*(a_.*v_)^m_*(b_.*v_+c_.*v_^2),x_Symbol] :=
  1/a*Int[u*(a*v)^(m+1)*(b+c*v),x] /;
FreeQ[{a,b,c},x] && LeQ[m,-1]
```

```
Int[u_.*(a_+b_.*v_)^m_*(A_.+B_.*v_+C_.*v_^2),x_Symbol] :=
  1/b^2*Int[u*(a+b*v)^(m+1)*Simp[b*B-a*C+b*C*v,x],x] /;
FreeQ[{a,b,A,B,C},x] && EqQ[A*b^2-a*b*B+a^2*C,0] && LeQ[m,-1]
```

10: $\int u\,(a+b\,x^n)^m\,(c+d\,x^{-n})^p\,dx$ when $a\,c-b\,d=0\,\wedge\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $a\,c-b\,d=0\,\wedge\,p\in\mathbb{Z}$, then $(c+d\,x^{-n})^p=\left(\frac{d}{a}\right)^p\frac{(a+b\,x^n)^p}{x^{n\,p}}$

Rule: If $a\,c-b\,d=0\,\wedge\,p\in\mathbb{Z}$, then

$$\int u\,(a+b\,x^n)^m\,(c+d\,x^{-n})^p\,dx\;\longrightarrow\;\left(\frac{d}{a}\right)^p\int\frac{u\,(a+b\,x^n)^{m+p}}{x^{n\,p}}\,dx$$

Program code:

```
Int[u_.*(a_+b_.*x_^n_.)^m_.*(c_+d_.*x_^q_.)^p_.,x_Symbol] :=
  (d/a)^p*Int[u*(a+b*x^n)^(m+p)/x^(n*p),x] /;
FreeQ[{a,b,c,d,m,n},x] && EqQ[q,-n] && IntegerQ[p] && EqQ[a*c-b*d,0] && Not[IntegerQ[m] && NegQ[n]]
```

11: $\int u\,(a+b\,x^n)^m\,(c+d\,x^{2\,n})^{-m}\,dx$ when $b^2\,c+a^2\,d==0\,\land\,a>0\,\land\,d<0$

Derivation: Algebraic simplification

Basis: If $b^2\,c+a^2\,d==0\,\land\,a>0\,\land\,d<0$, then $(a+b\,z)^m\,(c+d\,z^2)^{-m}==\left(-\frac{b^2}{d}\right)^m\,(a-b\,z)^{-m}$

Rule: If $b^2\,c+a^2\,d==0\,\land\,a>0\,\land\,d<0$, then

$$\int u\,(a+b\,x^n)^m\,(c+d\,x^{2\,n})^{-m}\,dx\;\longrightarrow\;\left(-\frac{b^2}{d}\right)^m\int u\,(a-b\,x^n)^{-m}\,dx$$

Program code:

```
Int[u_.*(a_+b_.*x_^n_.)^m_.*(c_+d_.*x_^j_)^p_.,x_Symbol] :=
  (-b^2/d)^m*Int[u*(a-b*x^n)^(-m),x] /;
FreeQ[{a,b,c,d,m,n,p},x] && EqQ[j,2*n] && EqQ[p,-m] && EqQ[b^2*c+a^2*d,0] && GtQ[a,0] && LtQ[d,0]
```

12: $\int u\,(a+b\,x^n+c\,x^{2\,n})^p\,dx$ when $b^2-4\,a\,c==0\,\land\,p\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b^2-4\,a\,c=0$, then $a+b\,z+c\,z^2==\frac{1}{c}\left(\frac{b}{2}+c\,z\right)^2$

Basis: If $b^2 - 4\,a\,c == 0$, then $a + b\,z + c\,z^2 == \left(\sqrt{a} + \frac{b\,z}{2\,\sqrt{a}}\right)^2$

Rule: If $b^2 - 4\,a\,c == 0 \land p \in \mathbb{Z}$, then

$$\int u\,\left(a + b\,x^n + c\,x^{2\,n}\right)^p\,dx \,\longrightarrow\, \frac{1}{c^p}\int u\,\left(\frac{b}{2} + c\,x^n\right)^{2\,p}\,dx$$

Program code:

```
Int[u_.*(a_+b_.*x_+c_.*x_^2)^p_.,x_Symbol] :=
  Int[u*Cancel[(b/2+c*x)^(2*p)/c^p],x] /;
FreeQ[{a,b,c},x] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

```
Int[u_.*(a_+b_.*x_^n_+c_.*x_^n2_.)^p_.,x_Symbol] :=
  1/c^p*Int[u*(b/2+c*x^n)^(2*p),x] /;
FreeQ[{a,b,c,n},x] && EqQ[n2,2*n] && EqQ[b^2-4*a*c,0] && IntegerQ[p]
```

13. $\int Q_r[x]\, F[P_q[x]]\, dx$

1: $\int \frac{P_p[x]}{Q_q[x]}\, dx$ when $p == q-1 \land P_p[x] == \frac{P_p[x,p]}{q\, Q_q[x,q]}\, \partial_x Q_q[x]$

Derivation: Reciprocal integration rule

Rule: If $p == q-1 \land P_p[x] == \frac{P_p[x,p]}{q\, Q_q[x,q]}\, \partial_x Q_q[x]$, then

$$\int \frac{P_p[x]}{Q_q[x]}\, dx \longrightarrow \frac{P_p[x,\, p]}{q\, Q_q[x,\, q]} \int \frac{\partial_x Q_q[x]}{Q_q[x]}\, dx \longrightarrow \frac{P_p[x,\, p]\, \text{Log}[Q_q[x]]}{q\, Q_q[x,\, q]}$$

Program code:

```
Int[Pp_/Qq_,x_Symbol] :=
  With[{p=Expon[Pp,x],q=Expon[Qq,x]},
  Coeff[Pp,x,p]*Log[RemoveContent[Qq,x]]/(q*Coeff[Qq,x,q])/;
 EqQ[p,q-1] && EqQ[Pp,Simplify[Coeff[Pp,x,p]/(q*Coeff[Qq,x,q])*D[Qq,x]]]] /;
PolyQ[Pp,x] && PolyQ[Qq,x]
```

2: $\int P_p[x]\, Q_q[x]^m\, dx$ when $m \neq -1 \wedge p + m q + 1 \neq 0 \wedge (p + m q + 1)\, Q_q[x, q]\, P_p[x] == P_p[x, p]\, x^{p-q}\, ((p - q + 1)\, Q_q[x] + (m + 1)\, x\, \partial_x Q_q[x])$

Derivation: Derivative divides

Basis: $x^{p-q}\, Q_q[x]^m\, ((p - q + 1)\, Q_q[x] + (m + 1)\, x\, \partial_x Q_q[x]) == \partial_x \left(x^{p-q+1}\, Q_q[x]^{m+1}\right)$

Note: The degree of the polynomial $x^{p-q}\, ((p - q + 1)\, Q_q[x] + (m + 1)\, x\, \partial_x Q_q[x])$ is p and the leading coefficient is $(p + m q + 1)\, Q_q[x, q]$.

Rule: If $m \neq -1 \wedge p + m q + 1 \neq 0 \wedge$
$(p + m q + 1)\, Q_q[x, q]\, P_p[x] == P_p[x, p]\, x^{p-q}\, ((p - q + 1)\, Q_q[x] + (m + 1)\, x\, \partial_x Q_q[x])$, then

$$\int P_p[x]\, Q_q[x]^m\, dx \longrightarrow \frac{P_p[x, p]}{(p + m q + 1)\, Q_q[x, q]} \int x^{p-q}\, Q_q[x]^m\, ((p - q + 1)\, Q_q[x] + (m + 1)\, x\, \partial_x Q_q[x])\, dx \longrightarrow \frac{P_p[x, p]\, x^{p-q+1}\, Q_q[x]^{m+1}}{(p + m q + 1)\, Q_q[x, q]}$$

Program code:

```
Int[Pp_*Qq_^m_.,x_Symbol] :=
  With[{p=Expon[Pp,x],q=Expon[Qq,x]},
  Coeff[Pp,x,p]*x^(p-q+1)*Qq^(m+1)/((p+m*q+1)*Coeff[Qq,x,q]) /;
 NeQ[p+m*q+1,0] && EqQ[(p+m*q+1)*Coeff[Qq,x,q]*Pp,Coeff[Pp,x,p]*x^(p-q)*((p-q+1)*Qq+(m+1)*x*D[Qq,x])]] /;
FreeQ[m,x] && PolyQ[Pp,x] && PolyQ[Qq,x] && NeQ[m,-1]
```

```
Int[x_^m_.*(a1_+b1_.*x_^n_.)^p_*(a2_+b2_.*x_^n_.)^p_,x_Symbol] :=
  (a1+b1*x^n)^(p+1)*(a2+b2*x^n)^(p+1)/(2*b1*b2*n*(p+1)) /;
FreeQ[{a1,b1,a2,b2,m,n,p},x] && EqQ[a2*b1+a1*b2,0] && EqQ[m-2*n+1,0] && NeQ[p,-1]
```

3: $\int P_p[x]\, Q_q[x]^m\, R_r[x]^n\, dx$ when

$m \neq -1 \land n \neq -1 \land p+m q+n r+1 \neq 0 \land$
$(p+m q+n r+1)\, Q_q[x,q]\, R_r[x,r]\, P_p[x] == P_p[x,p]\, x^{p-q-r}\, ((p-q-r+1)\, Q_q[x]\, R_r[x] + (m+1)\, x\, R_r[x]\, \partial_x Q_q[x] + (n+1)\, x\, Q_q[x]\, \partial_x R_r[x])$

Derivation: Derivative divides

Basis: $x^{p-q-r}\, Q_q[x]^m\, R_r[x]^n\, \left((p-q-r+1)\, Q_q[x]\, R_r[x] + (m+1)\, x\, R_r[x]\, \partial_x Q_q[x] + (n+1)\, x\, Q_q[x]\, \partial_x R_r[x]\right) == \partial_x \left(x^{p-q-r+1}\, Q_q[x]^{m+1}\, R_r[x]^{n+1}\right)$

Note: The degree of the polynomial $x^{p-q-r}\, \left((p-q-r+1)\, Q_q[x]\, R_r[x] + (m+1)\, x\, R_r[x]\, \partial_x Q_q[x] + (n+1)\, x\, Q_q[x]\, \partial_x R_r[x]\right)$ is p and the leading coefficient is $(p+m q+n r+1)\, Q_q[x,q]\, R_r[x,r]$.

Rule: If
$m \neq -1 \land n \neq -1 \land p+m q+n r+1 \neq 0 \land (p+m q+n r+1)\, Q_q[x,q]\, R_r[x,r]\, P_p[x] ==$,
$P_p[x,p]\, x^{p-q-r}\, ((p-q-r+1)\, Q_q[x]\, R_r[x] + (m+1)\, x\, R_r[x]\, \partial_x Q_q[x] + (n+1)\, x\, Q_q[x]\, \partial_x R_r[x])$
then

$$\int P_p[x]\, Q_q[x]^m\, R_r[x]^n\, dx \longrightarrow$$

$$\frac{P_p[x,p]}{(p+m q+n r+1)\, Q_q[x,q]\, R_r[x,r]} \int x^{p-q-r}\, Q_q[x]^m\, R_r[x]^n\, ((p-q-r+1)\, Q_q[x]\, R_r[x] + (m+1)\, x\, R_r[x]\, \partial_x Q_q[x] + (n+1)\, x\, Q_q[x]\, \partial_x R_r[x])\, dx \longrightarrow$$

$$\frac{P_p[x,p]\, x^{p-q-r+1}\, Q_q[x]^{m+1}\, R_r[x]^{n+1}}{(p+m q+n r+1)\, Q_q[x,q]\, R_r[x,r]}$$

Program code:

```
Int[Pp_*Qq_^m_.*Rr_^n_.,x_Symbol] :=
  With[{p=Expon[Pp,x],q=Expon[Qq,x],r=Expon[Rr,x]},
  Coeff[Pp,x,p]*x^(p-q-r+1)*Qq^(m+1)*Rr^(n+1)/((p+m*q+n*r+1)*Coeff[Qq,x,q]*Coeff[Rr,x,r]) /;
 NeQ[p+m*q+n*r+1,0] &&
 EqQ[(p+m*q+n*r+1)*Coeff[Qq,x,q]*Coeff[Rr,x,r]*Pp,Coeff[Pp,x,p]*x^(p-q-r)*((p-q-r+1)*Qq*Rr+(m+1)*x*Rr*D[Qq,x]+(n+1)*x*Qq*D[Rr,x])]] /;
FreeQ[{m,n},x] && PolyQ[Pp,x] && PolyQ[Qq,x] && PolyQ[Rr,x] && NeQ[m,-1] && NeQ[n,-1]
```

4: $\int Q_r[x]\,\left(a+b\,P_q[x]^n\right)^p\,dx$ when $\frac{Q_r[x]}{\partial_x P_q[x]} == \frac{Q_r[x,r]}{q\,P_q[x,q]}$

Derivation: Integration by substitution (derivative divides)

Basis: If $\frac{Q_r[x]}{\partial_x P_q[x]} == \frac{Q_r[x,r]}{q\,P_q[x,q]}$, then $F[P_q[x]]\,Q_r[x] == \frac{Q_r[x,r]}{q\,P_q[x,q]}\,\text{Subst}[F[x],\,x,\,P_q[x]]\,\partial_x P_q[x]$

Rule: If $\frac{Q_r[x]}{\partial_x P_q[x]} == \frac{Q_r[x,r]}{q\,P_q[x,q]}$, then

$$\int Q_r[x]\,\left(a+b\,P_q[x]^n\right)^p\,dx \longrightarrow \frac{Q_r[x,\,r]}{q\,P_q[x,\,q]}\,\text{Subst}\left[\int\left(a+b\,x^n\right)^p\,dx,\,x,\,P_q[x]\right]$$

Program code:

```
Int[Qr_*(a_.+b_.*Pq_^n_.)^p_.,x_Symbol] :=
  With[{q=Expon[Pq,x],r=Expon[Qr,x]},
  Coeff[Qr,x,r]/(q*Coeff[Pq,x,q])*Subst[Int[(a+b*x^n)^p,x],x,Pq] /;
 EqQ[r,q-1] && EqQ[Coeff[Qr,x,r]*D[Pq,x],q*Coeff[Pq,x,q]*Qr]] /;
FreeQ[{a,b,n,p},x] && PolyQ[Pq,x] && PolyQ[Qr,x]
```

5: $\int Q_r[x]\left(a+b\,P_q[x]^n+c\,P_q[x]^{2n}\right)^p\,dx$ when $\frac{Q_r[x]}{\partial_x P_q[x]}==\frac{Q_r[x,r]}{q\,P_q[x,q]}$

Derivation: Integration by substitution (derivative divides)

Basis: If $\frac{Q_r[x]}{\partial_x P_q[x]}==\frac{Q_r[x,r]}{q\,P_q[x,q]}$, then $F[P_q[x]]\,Q_r[x]==\frac{Q_r[x,r]}{q\,P_q[x,q]}\,\text{Subst}[F[x],x,P_q[x]]\,\partial_x P_q[x]$

Rule: If $\frac{Q_r[x]}{\partial_x P_q[x]}==\frac{Q_r[x,r]}{q\,P_q[x,q]}$, then

$$\int Q_r[x]\left(a+b\,P_q[x]^n+c\,P_q[x]^{2n}\right)^p\,dx\,\longrightarrow\,\frac{Q_r[x,r]}{q\,P_q[x,q]}\,\text{Subst}\left[\int\left(a+b\,x^n+c\,x^{2n}\right)^p\,dx,\,x,\,P_q[x]\right]$$

Program code:

```
Int[Qr_*(a_.+b_.*Pq_^n_.+c_.*Pq_^n2_.)^p_.,x_Symbol] :=
  Module[{q=Expon[Pq,x],r=Expon[Qr,x]},
  Coeff[Qr,x,r]/(q*Coeff[Pq,x,q])*Subst[Int[(a+b*x^n+c*x^(2*n))^p,x],x,Pq] /;
 EqQ[r,q-1] && EqQ[Coeff[Qr,x,r]*D[Pq,x],q*Coeff[Pq,x,q]*Qr]] /;
FreeQ[{a,b,c,n,p},x] && EqQ[n2,2*n] && PolyQ[Pq,x] && PolyQ[Qr,x]
```

14. $\int u\,(a\,x^p+b\,x^q+\cdots)^m\,dx$ when $m\in\mathbb{Z}$

1: $\int u\,(a\,x^p+b\,x^q)^m\,dx$ when $m\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: $a\,x^p+b\,x^q==x^p\,\left(a+b\,x^{q-p}\right)$

Rule: If $m\in\mathbb{Z}$, then

$$\int u\,(a\,x^p+b\,x^q)^m\,dx\;\longrightarrow\;\int u\,x^{m\,p}\,\left(a+b\,x^{q-p}\right)^m\,dx$$

Program code:

```
Int[u_.*(a_.*x_^p_.+b_.*x_^q_.)^m_.,x_Symbol] :=
  Int[u*x^(m*p)*(a+b*x^(q-p))^m,x] /;
FreeQ[{a,b,p,q},x] && IntegerQ[m] && PosQ[q-p]
```

2: $\int u \left(a x^p + b x^q + c x^r\right)^m dx$ when $m \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $a x^p + b x^q + c x^r == x^p \left(a + b x^{q-p} + c x^{r-p}\right)$

Rule: If $m \in \mathbb{Z}$, then

$$\int u \left(a x^p + b x^q + c x^r\right)^m dx \longrightarrow \int u\, x^{m p} \left(a + b x^{q-p} + c x^{r-p}\right)^m dx$$

Program code:

```
Int[u_.*(a_.*x_^p_.+b_.*x_^q_.+c_.*x_^r_.)^m_.,x_Symbol] :=
  Int[u*x^(m*p)*(a+b*x^(q-p)+c*x^(r-p))^m,x] /;
FreeQ[{a,b,c,p,q,r},x] && IntegerQ[m] && PosQ[q-p] && PosQ[r-p]
```

15. $\int u\, P[x]^p\, Q[x]^q \, dx$ when PolyGCD[P[x], Q[x], x] $\neq 1$

1. $\int P[x]^p\, Q[x]^q \, dx$ when $p + q == 0 \land$ PolynomialRemainder[P[x], Q[x], x] $== 0$

1: $\int P[x]^p\, Q[x]^q \, dx$ when $p + q == 0 \land$ PolynomialRemainder[P[x], Q[x], x] $== 0 \land p \in \mathbb{Z}$

Derivation: Algebraic simplification

Rule: If $p + q == 0 \land$ PolynomialRemainder[P[x], Q[x], x] $== 0 \land p \in \mathbb{Z}$, then

$$\int P[x]^p\, Q[x]^q \, dx \longrightarrow \int \text{PolynomialQuotient}[P[x], Q[x], x]^p \, dx$$

Program code:

```
Int[P_^p_.*Q_^q_.,x_Symbol] :=
  Int[PolynomialQuotient[P,Q,x]^p,x] /;
FreeQ[{p,q},x] && PolyQ[P,x] && PolyQ[Q,x] && EqQ[p+q,0] && EqQ[PolynomialRemainder[P,Q,x],0] && IntegerQ[p]
```

2: $\int P[x]^p\, Q[x]^q\, dx$ when $p + q == 0 \land \text{PolynomialRemainder}[P[x], Q[x], x] == 0 \land p \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Rule: If $p + q == 0 \land \text{PolynomialRemainder}[P[x], Q[x], x] == 0 \land p \notin \mathbb{Z}$, let $R[x] \to \text{PolynomialQuotient}[P[x], Q[x], x]$, then

$$\int P[x]^p\, Q[x]^q\, dx \longrightarrow \frac{P[x]^p\, Q[x]^q}{R[x]^p} \int R[x]^p\, dx$$

Program code:

```
(* Int[P_^p_.*Q_^q_.,x_Symbol] :=
  With[{R=PolynomialQuotient[P,Q,x]},
  P^p*Q^q/R^p*Int[R^p,x]] /;
FreeQ[{p,q},x] && PolyQ[P,x] && PolyQ[Q,x] && EqQ[p+q,0] && EqQ[PolynomialRemainder[P,Q,x],0] && Not[IntegerQ[p]] *)
```

2: $\int u\, P[x]^p\, Q[x]^q\, dx$ when $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^- \land \text{PolyGCD}[P[x], Q[x], x] \neq 1$

Derivation: Algebraic simplification

Rule: If $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^-$, let gcd $= \text{PolyGCD}[P[x], Q[x], x]$, if gcd $\neq 1$, then

$$\int u\, P[x]^p\, Q[x]^q\, dx \rightarrow \int u\, \text{gcd}^{p+q}\, \text{PolynomialQuotient}[P[x], \text{gcd}, x]^p\, \text{PolynomialQuotient}[Q[x], \text{gcd}, x]^q\, dx$$

Program code:

```
Int[u_.*P_^p_*Q_^q_,x_Symbol] :=
  Module[{gcd=PolyGCD[P,Q,x]},
  Int[u*gcd^(p+q)*PolynomialQuotient[P,gcd,x]^p*PolynomialQuotient[Q,gcd,x]^q,x] /;
 NeQ[gcd,1]] /;
IGtQ[p,0] && ILtQ[q,0] && PolyQ[P,x] && PolyQ[Q,x]
```

```
Int[u_.*P_*Q_^q_,x_Symbol] :=
  Module[{gcd=PolyGCD[P,Q,x]},
  Int[u*gcd^(q+1)*PolynomialQuotient[P,gcd,x]*PolynomialQuotient[Q,gcd,x]^q,x] /;
 NeQ[gcd,1]] /;
ILtQ[q,0] && PolyQ[P,x] && PolyQ[Q,x]
```