Rules for integrands of the form $(a + b\,\mathrm{Sec}[c + d\,x])^n$

1. $\int (b\,\mathrm{Sec}[c + d\,x])^n \, dx$

   1. $\int (b\,\mathrm{Sec}[c + d\,x])^n \, dx$ when $n > 1$

      1: $\int \mathrm{Sec}[c + d\,x]^n \, dx$ when $\frac{n}{2} \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If $\frac{n}{2} \in \mathbb{Z}$, then $\mathrm{Sec}[c + d\,x]^n == \frac{1}{d} \left(1 + \mathrm{Tan}[c + d\,x]^2\right)^{\frac{n}{2}-1} \partial_x \mathrm{Tan}[c + d\,x]$

Rule: If $\frac{n}{2} \in \mathbb{Z}^+$, then

$$\int \mathrm{Sec}[c + d\,x]^n \, dx \longrightarrow \frac{1}{d} \,\mathrm{Subst}\left[\int \left(1 + x^2\right)^{\frac{n}{2}-1} dx,\ x,\ \mathrm{Tan}[c + d\,x]\right]$$

Program code:

```
Int[csc[c_.+d_.*x_]^n_,x_Symbol] :=
  -1/d*Subst[Int[ExpandIntegrand[(1+x^2)^(n/2-1),x],x],x,Cot[c+d*x]] /;
FreeQ[{c,d},x] && IGtQ[n/2,0]
```

2: $\int (b \operatorname{Sec}[c+d\,x])^n \,dx$ when $n > 1$

Reference: CRC 313

Reference: CRC 309

Derivation: Secant recurrence 3a with $A \to 0$, $B \to a$, $C \to d$, $m \to m-1$, $n \to -1$

Rule: If $n > 1$, then

$$\int (b \operatorname{Sec}[c+d\,x])^n \,dx \to \frac{b \operatorname{Sin}[c+d\,x] \, (b \operatorname{Sec}[c+d\,x])^{n-1}}{d\,(n-1)} + \frac{b^2\,(n-2)}{n-1} \int (b \operatorname{Sec}[c+d\,x])^{n-2} \,dx$$

Program code:

```
Int[(b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  -b*Cos[c+d*x]*(b*Csc[c+d*x])^(n-1)/(d*(n-1)) +
  b^2*(n-2)/(n-1)*Int[(b*Csc[c+d*x])^(n-2),x] /;
FreeQ[{b,c,d},x] && GtQ[n,1] && IntegerQ[2*n]
```

2: $\int (b \operatorname{Sec}[c+d x])^n \, dx$ when $n < -1$

Reference: CRC 305

Reference: CRC 299

Derivation: Secant recurrence 1a with $A \to 1$, $B \to 0$, $C \to 0$, $n \to 0$

Rule: If $n < -1$, then

$$\int (b \operatorname{Sec}[c+d x])^n \, dx \longrightarrow -\frac{\operatorname{Sin}[c+d x] \, (b \operatorname{Sec}[c+d x])^{n+1}}{b d n} + \frac{(n+1)}{b^2 n} \int (b \operatorname{Sec}[c+d x])^{n+2} \, dx$$

Program code:

```
Int[(b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  Cos[c+d*x]*(b*Csc[c+d*x])^(n+1)/(b*d*n) +
  (n+1)/(b^2*n)*Int[(b*Csc[c+d*x])^(n+2),x] /;
FreeQ[{b,c,d},x] && LtQ[n,-1] && IntegerQ[2*n]
```

3: $\int \operatorname{Sec}[c+d x] \, dx$

Reference: G&R 2.526.9, CRC 294, A&S 4.3.117

Reference: G&R 2.526.1, CRC 295, A&S 4.3.116

Derivation: Integration by substitution

Basis: $\operatorname{Sec}[c+d x] == \frac{1}{d} \operatorname{Subst}\left[\frac{1}{1-x^2}, x, \operatorname{Sin}[c+d x]\right] \partial_x \operatorname{Sin}[c+d x]$

Rule:

$$\int \text{Sec}[c + d\,x]\,dx \longrightarrow \frac{\text{ArcTanh}[\text{Sin}[c + d\,x]]}{d}$$

Program code:

```
Int[csc[c_.+d_.*x_],x_Symbol] :=
(* -ArcCoth[Cos[c+d*x]]/d /; *)
  -ArcTanh[Cos[c+d*x]]/d /;
FreeQ[{c,d},x]
```

x: $\int \frac{1}{\text{Sec}[c + d\,x]}\,dx$

Note: This rule not necessary since *Mathematica* automatically simplifies $\frac{1}{\text{Sec}[z]}$ to $\text{Cos}[z]$.

Rule:

$$\int \frac{1}{\text{Sec}[c + d\,x]}\,dx \longrightarrow \int \text{Cos}[c + d\,x]\,dx \longrightarrow \frac{\text{Sin}[c + d\,x]}{d}$$

Program code:

```
(* Int[1/csc[c_.+d_.*x_],x_Symbol] :=
  -Cos[c+d*x]/d /;
FreeQ[{c,d},x] *)
```

4: $\int (b\,\text{Sec}[c+d\,x])^n\,dx$ when $n^2 == \frac{1}{4}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\,((b\,\text{Sec}[c+d\,x])^n\,(\text{Cos}[c+d\,x])^n) == 0$

Rule: If $n^2 == \frac{1}{4}$, then

$$\int (b\,\text{Sec}[c+d\,x])^n\,dx \longrightarrow (b\,\text{Sec}[c+d\,x])^n\,(\text{Cos}[c+d\,x])^n \int \frac{1}{\text{Cos}[c+d\,x]^n}\,dx$$

Program code:

```
Int[(b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  (b*Csc[c+d*x])^n*Sin[c+d*x]^n*Int[1/Sin[c+d*x]^n,x] /;
FreeQ[{b,c,d},x] && EqQ[n^2,1/4]
```

5: $\int (b\,\mathrm{Sec}[c+d\,x])^n\,dx$ when $n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x\left((b\,\mathrm{Sec}[c+d\,x])^n\,(\mathrm{Cos}[c+d\,x])^n\right) == 0$

Note: Decrementing the exponents in the piecewise constant factor results in canceling out the cosine factor introduced when integrating the power of the cosine.

Rule: If $2\,n \notin \mathbb{Z}$, then

$$\int (b\,\mathrm{Sec}[c+d\,x])^n\,dx \longrightarrow (b\,\mathrm{Sec}[c+d\,x])^{n-1}\left(\frac{\mathrm{Cos}[c+d\,x]}{b}\right)^{n-1}\int \frac{1}{\left(\frac{\mathrm{Cos}[c+d\,x]}{b}\right)^n}\,dx$$

Program code:

```
Int[(b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  (b*Csc[c+d*x])^(n-1)*((Sin[c+d*x]/b)^(n-1)*Int[1/(Sin[c+d*x]/b)^n,x]) /;
FreeQ[{b,c,d,n},x] && Not[IntegerQ[n]]
```

2: $\int (a + b \operatorname{Sec}[c + d x])^2 \, dx$

Derivation: Algebraic expansion

Basis: $(a + b z)^2 == a^2 + 2 a b z + b^2 z^2$

Rule:

$$\int (a + b \operatorname{Sec}[c + d x])^2 \, dx \to a^2 x + 2 a b \int \operatorname{Sec}[c + d x] \, dx + b^2 \int \operatorname{Sec}[c + d x]^2 \, dx$$

Program code:

```
Int[(a_+b_.*csc[c_.+d_.*x_])^2,x_Symbol] :=
  a^2*x + 2*a*b*Int[Csc[c+d*x],x] + b^2*Int[Csc[c+d*x]^2,x] /;
FreeQ[{a,b,c,d},x]
```

3. $\int (a + b \operatorname{Sec}[c + d x])^n \, dx$ when $a^2 - b^2 == 0$

1. $\int (a + b \operatorname{Sec}[c + d x])^n \, dx$ when $a^2 - b^2 == 0 \land 2 n \in \mathbb{Z}$

1. $\int (a + b \operatorname{Sec}[c + d x])^n \, dx$ when $a^2 - b^2 == 0 \land 2 n \in \mathbb{Z}^+$

1: $\int \sqrt{a + b \operatorname{Sec}[c + d x]} \, dx$ when $a^2 - b^2 == 0$

Author: Martin Welz on 24 June 2011

Derivation: Integration by substitution

Basis: If $a^2 - b^2 == 0$, then $\sqrt{a + b \operatorname{Sec}[c + d x]} == \frac{2 b}{d} \operatorname{Subst}\left[\frac{1}{a + x^2}, x, \frac{b \operatorname{Tan}[c + d x]}{\sqrt{a + b \operatorname{Sec}[c + d x]}}\right] \partial_x \frac{b \operatorname{Tan}[c + d x]}{\sqrt{a + b \operatorname{Sec}[c + d x]}}$

Rule: If $a^2 - b^2 == 0$, then

$$\int \sqrt{a+b\,\text{Sec}[c+d\,x]}\,dx \to \frac{2\,b}{d}\,\text{Subst}\left[\int \frac{1}{a+x^2}\,dx,\,x,\,\frac{b\,\text{Tan}[c+d\,x]}{\sqrt{a+b\,\text{Sec}[c+d\,x]}}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*csc[c_.+d_.*x_]],x_Symbol] :=
  -2*b/d*Subst[Int[1/(a+x^2),x],x,b*Cot[c+d*x]/Sqrt[a+b*Csc[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0]
```

2: $\int (a+b\,\text{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2=0 \land n>1 \land 2\,n\in\mathbb{Z}$

Derivation: Symmetric secant recurrence 1b with $A\to a,\ B\to b,\ m\to 0,\ n\to n-1$

Rule: If $a^2-b^2=0 \land n>1 \land 2\,n\in\mathbb{Z}$, then

$$\int (a+b\,\text{Sec}[c+d\,x])^n\,dx \to$$

$$\frac{b^2\,\text{Tan}[c+d\,x]\,(a+b\,\text{Sec}[c+d\,x])^{n-2}}{d\,(n-1)}+\frac{a}{n-1}\int (a+b\,\text{Sec}[c+d\,x])^{n-2}\,(a\,(n-1)+b\,(3\,n-4)\,\text{Sec}[c+d\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  -b^2*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n-2)/(d*(n-1)) +
  a/(n-1)*Int[(a+b*Csc[c+d*x])^(n-2)*(a*(n-1)+b*(3*n-4)*Csc[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0] && GtQ[n,1] && IntegerQ[2*n]
```

2. $\int (a + b \operatorname{Sec}[c + d x])^n \, dx$ when $a^2 - b^2 == 0 \land 2n \in \mathbb{Z}^-$

1: $\int \frac{1}{\sqrt{a + b \operatorname{Sec}[c + d x]}} \, dx$ when $a^2 - b^2 == 0$

Author: Martin on sci.math.symbolic on 10 March 2011

Derivation: Algebraic expansion

Basis: $\frac{1}{\sqrt{a+b z}} == \frac{\sqrt{a+b z}}{a} - \frac{b z}{a \sqrt{a+b z}}$

Rule: If $a^2 - b^2 == 0$, then

$$\int \frac{1}{\sqrt{a + b \operatorname{Sec}[c + d x]}} \, dx \to \frac{1}{a} \int \sqrt{a + b \operatorname{Sec}[c + d x]} \, dx - \frac{b}{a} \int \frac{\operatorname{Sec}[c + d x]}{\sqrt{a + b \operatorname{Sec}[c + d x]}} \, dx$$

Program code:

```
Int[1/Sqrt[a_+b_.*csc[c_.+d_.*x_]],x_Symbol] :=
  1/a*Int[Sqrt[a+b*Csc[c+d*x]],x] -
  b/a*Int[Csc[c+d*x]/Sqrt[a+b*Csc[c+d*x]],x] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0]
```

2: $\int (a + b \operatorname{Sec}[c + d x])^n \, dx$ when $a^2 - b^2 == 0 \land n \leq -1 \land 2n \in \mathbb{Z}$

Derivation: Symmetric secant recurrence 2b with $A \to 1, B \to 0, m \to 0$

Rule: If $a^2 - b^2 == 0 \land n \leq -1 \land 2n \in \mathbb{Z}$, then

$$\int (a + b \operatorname{Sec}[c + d x])^n \, dx \to$$

$$\frac{\mathrm{Tan}[c+d\,x]\,(a+b\,\mathrm{Sec}[c+d\,x])^n}{d\,(2\,n+1)}+\frac{1}{a^2\,(2\,n+1)}\int(a+b\,\mathrm{Sec}[c+d\,x])^{n+1}\,(a\,(2\,n+1)-b\,(n+1)\,\mathrm{Sec}[c+d\,x])\,dx$$

Program code:

```
Int[(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  -Cot[c+d*x]*(a+b*Csc[c+d*x])^n/(d*(2*n+1)) +
  1/(a^2*(2*n+1))*Int[(a+b*Csc[c+d*x])^(n+1)*(a*(2*n+1)-b*(n+1)*Csc[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0] && LeQ[n,-1] && IntegerQ[2*n]
```

2. $\int(a+b\,\mathrm{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2=0\land 2\,n\notin\mathbb{Z}$

1: $\int(a+b\,\mathrm{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2=0\land 2\,n\notin\mathbb{Z}\land a>0$

Derivation: Piecewise constant extraction and integration by substitution

Basis: If $a^2-b^2=0$, then $\partial_x\frac{\mathrm{Tan}[c+d\,x]}{\sqrt{1+\mathrm{Sec}[c+d\,x]}\,\sqrt{1-\mathrm{Sec}[c+d\,x]}}=0$

Basis: If $a^2-b^2=0\land a>0$, then $-\frac{\mathrm{Tan}[c+d\,x]}{\sqrt{1+\mathrm{Sec}[c+d\,x]}\,\sqrt{1-\mathrm{Sec}[c+d\,x]}}\,\frac{\mathrm{Tan}[c+d\,x]}{\sqrt{1+\frac{b}{a}\,\mathrm{Sec}[c+d\,x]}\,\sqrt{1-\frac{b}{a}\,\mathrm{Sec}[c+d\,x]}}=1$

Basis: $\mathrm{Tan}[c+d\,x]\,F[\mathrm{Sec}[c+d\,x]]=\frac{1}{d}\,\mathrm{Subst}\left[\frac{F[x]}{x},\,x,\,\mathrm{Sec}[c+d\,x]\right]\,\partial_x\,\mathrm{Sec}[c+d\,x]$

Rule: If $a^2-b^2=0\land 2\,n\notin\mathbb{Z}\land a>0$, then

$$\int(a+b\,\mathrm{Sec}[c+d\,x])^n\,dx\;\rightarrow\;a^n\int\left(1+\frac{b}{a}\,\mathrm{Sec}[c+d\,x]\right)^n\,dx\;\rightarrow$$

$$-\frac{a^n\,\mathrm{Tan}[c+d\,x]}{\sqrt{1+\mathrm{Sec}[c+d\,x]}\,\sqrt{1-\mathrm{Sec}[c+d\,x]}}\int\frac{\mathrm{Tan}[c+d\,x]\,\left(1+\frac{b}{a}\,\mathrm{Sec}[c+d\,x]\right)^{n-\frac{1}{2}}}{\sqrt{1-\frac{b}{a}\,\mathrm{Sec}[c+d\,x]}}\,dx\;\rightarrow$$

$$-\frac{a^n \operatorname{Tan}[c+d\,x]}{d\sqrt{1+\operatorname{Sec}[c+d\,x]}\,\sqrt{1-\operatorname{Sec}[c+d\,x]}}\operatorname{Subst}\left[\int\frac{\left(1+\frac{b\,x}{a}\right)^{n-\frac{1}{2}}}{x\sqrt{1-\frac{b\,x}{a}}}\,dx,\,x,\,\operatorname{Sec}[c+d\,x]\right]$$

Program code:

```
Int[(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  a^n*Cot[c+d*x]/(d*Sqrt[1+Csc[c+d*x]]*Sqrt[1-Csc[c+d*x]])*
    Subst[Int[(1+b*x/a)^(n-1/2)/(x*Sqrt[1-b*x/a]),x],x,Csc[c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[2*n]] && GtQ[a,0]
```

2: $\int (a+b\operatorname{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2==0\,\wedge\,2\,n\notin\mathbb{Z}\,\wedge\,a\ngtr 0$

Derivation: Piecewise constant extraction

Basis: If $\partial_x \frac{(a+b\operatorname{Sec}[c+d\,x])^n}{\left(1+\frac{b}{a}\operatorname{Sec}[c+d\,x]\right)^n}==0$

Rule: If $a^2-b^2==0\,\wedge\,2\,n\notin\mathbb{Z}\,\wedge\,a\ngtr 0$, then

$$\int (a+b\operatorname{Sec}[c+d\,x])^n\,dx\;\rightarrow\;\frac{a^{\operatorname{IntPart}[n]}\,(a+b\operatorname{Sec}[c+d\,x])^{\operatorname{FracPart}[n]}}{\left(1+\frac{b}{a}\operatorname{Sec}[c+d\,x]\right)^{\operatorname{FracPart}[n]}}\int\left(1+\frac{b}{a}\operatorname{Sec}[c+d\,x]\right)^n dx$$

Program code:

```
Int[(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  a^IntPart[n]*(a+b*Csc[c+d*x])^FracPart[n]/(1+b/a*Csc[c+d*x])^FracPart[n]*Int[(1+b/a*Csc[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[2*n]] && Not[GtQ[a,0]]
```

4. $\int (a + b \operatorname{Sec}[c + d x])^n \, dx$ when $a^2 - b^2 \neq 0 \land 2 n \in \mathbb{Z}$

   1. $\int (a + b \operatorname{Sec}[c + d x])^n \, dx$ when $a^2 - b^2 \neq 0 \land 2 n \in \mathbb{Z}^+$

      1: $\int \sqrt{a + b \operatorname{Sec}[c + d x]} \, dx$ when $a^2 - b^2 \neq 0$

Rule: If $a^2 - b^2 \neq 0$, then

$$\int \sqrt{a + b \operatorname{Sec}[c + d x]} \, dx \longrightarrow$$

$$-\frac{2 (a + b \operatorname{Sec}[c + d x])}{d \sqrt{a + b} \operatorname{Tan}[c + d x]} \sqrt{\frac{b (1 + \operatorname{Sec}[c + d x])}{a + b \operatorname{Sec}[c + d x]}} \sqrt{-\frac{b (1 - \operatorname{Sec}[c + d x])}{a + b \operatorname{Sec}[c + d x]}} \operatorname{EllipticPi}\left[\frac{a}{a + b}, \operatorname{ArcSin}\left[\frac{\sqrt{a + b}}{\sqrt{a + b \operatorname{Sec}[c + d x]}}\right], \frac{a - b}{a + b}\right]$$

Program code:

```
Int[Sqrt[a_+b_.*csc[c_.+d_.*x_]],x_Symbol] :=
  2*(a+b*Csc[c+d*x])/(d*Rt[a+b,2]*Cot[c+d*x])*Sqrt[b*(1+Csc[c+d*x])/(a+b*Csc[c+d*x])]*Sqrt[-b*(1-Csc[c+d*x])/(a+b*Csc[c+d*x])]*
    EllipticPi[a/(a+b),ArcSin[Rt[a+b,2]/Sqrt[a+b*Csc[c+d*x]]],(a-b)/(a+b)] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0]
```

2: $\int (a + b \operatorname{Sec}[c + d x])^{3/2} \, dx$ when $a^2 - b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $(a + b z)^{3/2} == a^2 \frac{1+z}{\sqrt{a+b z}} - \frac{z (a^2-2 a b-b^2 z)}{\sqrt{a+b z}}$

Rule: If $a^2 - b^2 \neq 0$, then

$$\int (a+b\,\text{Sec}[c+d\,x])^{3/2}\,dx \to \int \frac{a^2+b\,(2\,a-b)\,\text{Sec}[c+d\,x]}{\sqrt{a+b\,\text{Sec}[c+d\,x]}}\,dx + b^2\int \frac{\text{Sec}[c+d\,x]\,(1+\text{Sec}[c+d\,x])}{\sqrt{a+b\,\text{Sec}[c+d\,x]}}\,dx$$

Program code:

```
Int[(a_+b_.*csc[c_.+d_.*x_])^(3/2),x_Symbol] :=
  Int[(a^2+b*(2*a-b)*Csc[c+d*x])/Sqrt[a+b*Csc[c+d*x]],x] +
  b^2*Int[Csc[c+d*x]*(1+Csc[c+d*x])/Sqrt[a+b*Csc[c+d*x]],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0]
```

3: $\int (a+b\,\text{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2\neq 0\land n>2\land 2\,n\in\mathbb{Z}$

Derivation: Secant recurrence 1b with $A\to a^2,\ B\to 2\,a\,b,\ C\to b^2,\ m\to 0,\ n\to n-2$

Rule: If $a^2-b^2\neq 0\land n>2\land 2\,n\in\mathbb{Z}$, then

$$\int (a+b\,\text{Sec}[c+d\,x])^n\,dx \to$$
$$\frac{b^2\,\text{Tan}[c+d\,x]\,(a+b\,\text{Sec}[c+d\,x])^{n-2}}{d\,(n-1)}+$$
$$\frac{1}{n-1}\int (a+b\,\text{Sec}[c+d\,x])^{n-3}\,\left(a^3\,(n-1)+b\,\left(b^2\,(n-2)+3\,a^2\,(n-1)\right)\,\text{Sec}[c+d\,x]+a\,b^2\,(3\,n-4)\,\text{Sec}[c+d\,x]^2\right)\,dx$$

Program code:

```
Int[(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  -b^2*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n-2)/(d*(n-1)) +
  1/(n-1)*Int[(a+b*Csc[c+d*x])^(n-3)*
    Simp[a^3*(n-1)+(b*(b^2*(n-2)+3*a^2*(n-1)))*Csc[c+d*x]+(a*b^2*(3*n-4))*Csc[c+d*x]^2,x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && GtQ[n,2] && IntegerQ[2*n]
```

2. $\int (a + b \operatorname{Sec}[c + d x])^n \, dx$ when $a^2 - b^2 \neq 0 \land 2 n \in \mathbb{Z}^-$

1: $\int \frac{1}{a + b \operatorname{Sec}[c + d x]} \, dx$ when $a^2 - b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b z} == \frac{1}{a} - \frac{b z}{a (a+b z)}$

Rule: If $a^2 - b^2 \neq 0$, then

$$\int \frac{1}{a + b \operatorname{Sec}[c + d x]} \, dx \longrightarrow \frac{x}{a} - \frac{b}{a} \int \frac{\operatorname{Sec}[c + d x]}{a + b \operatorname{Sec}[c + d x]} \, dx \longrightarrow \frac{x}{a} - \frac{1}{a} \int \frac{1}{1 + \frac{a \operatorname{Cos}[c+d x]}{b}} \, dx$$

Program code:

```
Int[1/(a_+b_.*csc[c_.+d_.*x_]),x_Symbol] :=
  x/a - 1/a*Int[1/(1+a/b*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0]
```

2: $\int \frac{1}{\sqrt{a + b \operatorname{Sec}[c + d x]}} \, dx$ when $a^2 - b^2 \neq 0$

Rule: If $a^2 - b^2 \neq 0$, then

$$\int \frac{1}{\sqrt{a + b \operatorname{Sec}[c + d x]}} \, dx \longrightarrow$$

$$-\frac{2 \sqrt{a + b}}{a\, d \operatorname{Tan}[c + d x]} \sqrt{\frac{b\, (1 - \operatorname{Sec}[c + d x])}{a + b}} \sqrt{-\frac{b\, (1 + \operatorname{Sec}[c + d x])}{a - b}} \operatorname{EllipticPi}\left[\frac{a + b}{a}, \operatorname{ArcSin}\left[\frac{\sqrt{a + b \operatorname{Sec}[c + d x]}}{\sqrt{a + b}}\right], \frac{a + b}{a - b}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*csc[c_.+d_.*x_]],x_Symbol] :=
  2*Rt[a+b,2]/(a*d*Cot[c+d*x])*Sqrt[b*(1-Csc[c+d*x])/(a+b)]*Sqrt[-b*(1+Csc[c+d*x])/(a-b)]*
    EllipticPi[(a+b)/a,ArcSin[Sqrt[a+b*Csc[c+d*x]]/Rt[a+b,2]],(a+b)/(a-b)] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0]
```

3: $\int (a+b\,\text{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2\neq 0\land n<-1\land 2\,n\in\mathbb{Z}$

Derivation: Secant recurrence 2b with $A\to 1,\ B\to 0,\ C\to 0,\ m\to 0$

Rule: If $a^2-b^2\neq 0\ \land\ n<-1\ \land\ 2\,n\in\mathbb{Z}$, then

$$\int (a+b\,\text{Sec}[c+d\,x])^n\,dx\ \longrightarrow$$

$$-\frac{b^2\,\text{Tan}[c+d\,x]\,(a+b\,\text{Sec}[c+d\,x])^{n+1}}{a\,d\,(n+1)\,(a^2-b^2)}+$$

$$\frac{1}{a\,(n+1)\,(a^2-b^2)}\int (a+b\,\text{Sec}[c+d\,x])^{n+1}\,\left((a^2-b^2)\,(n+1)-a\,b\,(n+1)\,\text{Sec}[c+d\,x]+b^2\,(n+2)\,\text{Sec}[c+d\,x]^2\right)dx$$

Program code:

```
Int[(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  b^2*Cot[c+d*x]*(a+b*Csc[c+d*x])^(n+1)/(a*d*(n+1)*(a^2-b^2)) +
  1/(a*(n+1)*(a^2-b^2))*Int[(a+b*Csc[c+d*x])^(n+1)*Simp[(a^2-b^2)*(n+1)-a*b*(n+1)*Csc[c+d*x]+b^2*(n+2)*Csc[c+d*x]^2,x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && LtQ[n,-1] && IntegerQ[2*n]
```

X: $\int (a + b \operatorname{Sec}[c + d x])^n \, dx$ when $a^2 - b^2 \neq 0 \land 2 n \notin \mathbb{Z}$

Rule: If $a^2 - b^2 \neq 0 \land 2 n \notin \mathbb{Z}$, then

$$\int (a + b \operatorname{Sec}[c + d x])^n \, dx \longrightarrow \int (a + b \operatorname{Sec}[c + d x])^n \, dx$$

Program code:

```
Int[(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  Unintegrable[(a+b*Csc[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[a^2-b^2,0] && Not[IntegerQ[2*n]]
```