

Rules for integrands of the form $(a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x])$

1. $\int (a + b \sec[e + f x]) (d \sec[e + f x])^n (A + B \sec[e + f x]) dx$ when $A b - a B \neq 0$

1: $\int (a + b \sec[e + f x]) (d \sec[e + f x])^n (A + B \sec[e + f x]) dx$ when $A b - a B \neq 0 \wedge n \leq -1$

Derivation: Nondegenerate secant recurrence 1a with $A \rightarrow a A$, $B \rightarrow A b + a B$, $C \rightarrow b B$, $m \rightarrow 0$, $p \rightarrow 0$

Rule: If $A b - a B \neq 0 \wedge n \leq -1$, then

$$\int (a + b \sec[e + f x]) (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \rightarrow$$

$$- \frac{A a \tan[e + f x] (d \sec[e + f x])^n}{f n} + \frac{1}{d n} \int (d \sec[e + f x])^{n+1} (n (B a + A b) + (B b n + A a (n + 1)) \sec[e + f x]) dx$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A*a*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*n) +
  1/(d*n)*Int[(d*Csc[e+f*x])^(n+1)*Simp[n*(B*a+A*b)+(B*b*n+A*a*(n+1))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && LeQ[n,-1]
```

2: $\int (a + b \sec[e + f x]) (d \sec[e + f x])^n (A + B \sec[e + f x]) dx$ when $A b - a B \neq 0 \wedge n \neq -1$

Derivation: Nondegenerate secant recurrence 1b with $A \rightarrow a A$, $B \rightarrow A b + a B$, $C \rightarrow b B$, $m \rightarrow 0$, $p \rightarrow 0$

Rule: If $A b - a B \neq 0 \wedge n \neq -1$, then

$$\int (a + b \sec[e + f x]) (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \rightarrow$$

$$\frac{b B \tan[e + f x] (d \sec[e + f x])^n}{f (n + 1)} + \frac{1}{n + 1} \int (d \sec[e + f x])^n (A a (n + 1) + B b n + (A b + B a) (n + 1) \sec[e + f x]) dx$$

Program code:

```
Int[(a_+b_.*csc[e_+f_.*x_])*(d_.*csc[e_+f_.*x_])^n_.*(A_+B_.*csc[e_+f_.*x_]),x_Symbol] :=
  -b*B*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*(n+1)) +
  1/(n+1)*Int[(d*Csc[e+f*x])^n*Simp[A*a*(n+1)+B*b*n+(A*b+B*a)*(n+1)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && Not[LeQ[n,-1]]
```

2. $\int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx$ when $Ab - aB \neq 0$

1: $\int \frac{\sec[e+fx] (A+B \sec[e+fx])}{a+b \sec[e+fx]} dx$ when $Ab - aB \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+Bz}{a+bz} == \frac{B}{b} + \frac{Ab-aB}{b(a+bz)}$

Rule: If $Ab - aB \neq 0$, then

$$\int \frac{\sec[e+fx] (A+B \sec[e+fx])}{a+b \sec[e+fx]} dx \rightarrow \frac{B}{b} \int \sec[e+fx] dx + \frac{Ab-aB}{b} \int \frac{\sec[e+fx]}{a+b \sec[e+fx]} dx$$

Program code:

```
Int[csc[e_+f_.*x_]*(A_+B_.*csc[e_+f_.*x_])/(a_+b_.*csc[e_+f_.*x_]),x_Symbol] :=
  B/b*Int[Csc[e+fx],x] + (A*b-a*B)/b*Int[Csc[e+fx]/(a+b*Csc[e+fx]),x] /;
FreeQ[{a,b,e,f,A,B},x] && NeQ[A*b-a*B,0]
```

2. $\int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx$ when $Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0$

1: $\int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx$ when $Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge aBm + Ab(m+1) \neq 0$

Derivation: Singly degenerate secant recurrence 2a with $A \rightarrow -\frac{aBm}{b(m+1)}$, $n \rightarrow 0$, $p \rightarrow 0$

Derivation: Singly degenerate secant recurrence 2c with $A \rightarrow -\frac{aBm}{b(m+1)}$, $n \rightarrow 0$, $p \rightarrow 0$

Note: If $a^2 - b^2 \neq 0 \wedge aBm + Ab(m+1) \neq 0$, then $m+1 \neq 0$.

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge aBm + Ab(m+1) \neq 0$, then

$$\int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \rightarrow \frac{B \tan[e+fx] (a+b \sec[e+fx])^m}{f(m+1)}$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -B*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(m+1)) /;
FreeQ[{a,b,A,B,e,f,m},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && EqQ[a*B*m+A*b*(m+1),0]
```

$$2. \int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \text{ when } A b - a B \neq 0 \wedge a^2 - b^2 = 0 \wedge a B m + A b (m+1) \neq 0$$

$$1: \int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \text{ when } A b - a B \neq 0 \wedge a^2 - b^2 = 0 \wedge a B m + A b (m+1) \neq 0 \wedge m < -\frac{1}{2}$$

Derivation: Singly degenerate secant recurrence 2a with $n \rightarrow 0$, $p \rightarrow 0$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 = 0 \wedge a B m + A b (m+1) \neq 0 \wedge m < -\frac{1}{2}$, then

$$\int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \rightarrow$$

$$- \frac{(A b - a B) \tan[e+fx] (a+b \sec[e+fx])^m}{a f (2 m + 1)} + \frac{a B m + A b (m+1)}{a b (2 m + 1)} \int \sec[e+fx] (a+b \sec[e+fx])^{m+1} dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  (A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(a*f*(2*m+1)) +
  (a*B*m+A*b*(m+1))/(a*b*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1),x] /;
FreeQ[{a,b,A,B,e,f},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && NeQ[a*B*m+A*b*(m+1),0] && LtQ[m,-1/2]
```

2: $\int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 = 0 \wedge a B m + A b (m+1) \neq 0 \wedge m \neq -\frac{1}{2}$

Derivation: Singly degenerate secant recurrence 2c with $n \rightarrow 0$, $p \rightarrow 0$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 = 0 \wedge a B m + A b (m+1) \neq 0 \wedge m \neq -\frac{1}{2}$, then

$$\int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \rightarrow \frac{B \tan[e+fx] (a+b \sec[e+fx])^m}{f (m+1)} + \frac{a B m + A b (m+1)}{b (m+1)} \int \sec[e+fx] (a+b \sec[e+fx])^m dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -B*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(m+1)) +
  (a*B*m+A*b*(m+1))/(b*(m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m,x] /;
FreeQ[{a,b,A,B,e,f,m},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && NeQ[a*B*m+A*b*(m+1),0] && Not[LtQ[m,-1/2]]
```

$$3. \int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0$$

$$1: \int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m > 0$$

Reference: G&R 2.551.1 inverted

Derivation: Nondegenerate secant recurrence 1b with $A \rightarrow aA$, $B \rightarrow Ab + aB$, $C \rightarrow bB$, $m \rightarrow 0$, $n \rightarrow n - 1$, $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m > 0$, then

$$\int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \rightarrow \frac{B \tan[e+fx] (a+b \sec[e+fx])^m}{f(m+1)} + \frac{1}{m+1} \int \sec[e+fx] (a+b \sec[e+fx])^{m-1} (bBm + aC(m+1) + (aBm + Ab(m+1)) \sec[e+fx]) dx$$

Program code:

```
Int[csc[e_+f_.*x_]*(a_+b_.*csc[e_+f_.*x_])^m_*(A_+B_.*csc[e_+f_.*x_]),x_Symbol] :=
  -B*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(f*(m+1)) +
  1/(m+1)*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*Simp[b*B*m+a*A*(m+1)+(a*B*m+A*b*(m+1))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,A,B,e,f},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && GtQ[m,0]
```

$$2: \int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m < -1$$

Reference: G&R 2.551.1

Derivation: Nondegenerate secant recurrence 1a with $C \rightarrow 0$, $n \rightarrow 0$, $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m < -1$, then

$$\int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \rightarrow$$

$$\frac{(A b - a B) \tan[e + f x] (a + b \sec[e + f x])^{m+1}}{f (m+1) (a^2 - b^2)} + \frac{1}{(m+1) (a^2 - b^2)} \int \sec[e + f x] (a + b \sec[e + f x])^{m+1} ((a A - b B) (m+1) - (A b - a B) (m+2) \sec[e + f x]) dx$$

Program code:

```
Int[csc[e_+f_.*x_]*(a_+b_.*csc[e_+f_.*x_])^m_*(A_+B_.*csc[e_+f_.*x_]),x_Symbol] :=
  -(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(f*(m+1)*(a^2-b^2)) +
  1/((m+1)*(a^2-b^2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*Simp[(a*A-b*B)*(m+1)-(A*b-a*B)*(m+2)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,A,B,e,f},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

$$3. \int \frac{\sec[e + f x] (A + B \sec[e + f x])}{\sqrt{a + b \sec[e + f x]}} dx \text{ when } A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$$

$$1: \int \frac{\sec[e + f x] (A + B \sec[e + f x])}{\sqrt{a + b \sec[e + f x]}} dx \text{ when } a^2 - b^2 \neq 0 \wedge A^2 - B^2 = 0$$

Derivation: Piecewise constant extraction and integration by substitution

$$\text{Basis: } \partial_x \left(\frac{1}{\tan[e + f x]} \sqrt{\frac{b(1 - \sec[e + f x])}{a + b}} \sqrt{-\frac{b(1 + \sec[e + f x])}{a - b}} \right) = 0$$

$$\text{Basis: } \sec[e + f x] \tan[e + f x] F[\sec[e + f x]] = \frac{1}{f} \text{Subst}[F[x], x, \sec[e + f x]] \partial_x \sec[e + f x]$$

Rule: If $a^2 - b^2 \neq 0$, then

$$\int \frac{\sec[e + f x] (A + B \sec[e + f x])}{\sqrt{a + b \sec[e + f x]}} dx \rightarrow \frac{A b - a B}{b \tan[e + f x]}$$

$$\begin{aligned}
& \sqrt{\frac{b(1-\sec[e+fx])}{a+b}} \sqrt{-\frac{b(1+\sec[e+fx])}{a-b}} \int \frac{\sec[e+fx] \tan[e+fx] \sqrt{-\frac{bB}{aA-bB} - \frac{Ab \sec[e+fx]}{aA-bB}}}{\sqrt{a+b \sec[e+fx]} \sqrt{\frac{bB}{aA+bB} - \frac{Ab \sec[e+fx]}{aA+bB}}} dx \\
& \rightarrow \frac{Ab-aB}{bf \tan[e+fx]} \sqrt{\frac{b(1-\sec[e+fx])}{a+b}} \sqrt{-\frac{b(1+\sec[e+fx])}{a-b}} \text{Subst} \left[\int \frac{\sqrt{-\frac{bB}{aA-bB} - \frac{Abx}{aA-bB}}}{\sqrt{a+bx} \sqrt{\frac{bB}{aA+bB} - \frac{Abx}{aA+bB}}} dx, x, \sec[e+fx] \right] \\
& \rightarrow \frac{2(Ab-aB)}{b^2 f \tan[e+fx]} \sqrt{a + \frac{bB}{A}} \sqrt{\frac{b(1-\sec[e+fx])}{a+b}} \sqrt{-\frac{b(1+\sec[e+fx])}{a-b}} \text{EllipticE} \left[\text{ArcSin} \left[\frac{\sqrt{a+b \sec[e+fx]}}{\sqrt{a + \frac{bB}{A}}} \right], \frac{aA+bB}{aA-bB} \right]
\end{aligned}$$

Program code:

```

Int[csc[e_+f_.*x_]*(A_+B_.*csc[e_+f_.*x_])/Sqrt[a_+b_.*csc[e_+f_.*x_]],x_Symbol] :=
-2*(A*b-a*B)*Rt[a+b*B/A,2]*Sqrt[b*(1-Csc[e+f*x])/(a+b)]*Sqrt[-b*(1+Csc[e+f*x])/(a-b)]/(b^2*f*Cot[e+f*x])*
EllipticE[ArcSin[Sqrt[a+b*Csc[e+f*x]]/Rt[a+b*B/A,2]],(a*A+b*B)/(a*A-b*B)] /;
FreeQ[{a,b,e,f,A,B},x] && NeQ[a^2-b^2,0] && EqQ[A^2-B^2,0]

```

2: $\int \frac{\sec[e+fx] (A+B \sec[e+fx])}{\sqrt{a+b \sec[e+fx]}} dx$ when $a^2 - b^2 \neq 0 \wedge A^2 - B^2 \neq 0$

Derivation: Algebraic expansion

Basis: $A + Bz == A - B + B(1+z)$

Rule: If $a^2 - b^2 \neq 0 \wedge A^2 - B^2 \neq 0$, then

$$\int \frac{\sec[e+fx] (A+B \sec[e+fx])}{\sqrt{a+b \sec[e+fx]}} dx \rightarrow (A-B) \int \frac{\sec[e+fx]}{\sqrt{a+b \sec[e+fx]}} dx + B \int \frac{\sec[e+fx] (1+\sec[e+fx])}{\sqrt{a+b \sec[e+fx]}} dx$$

Program code:

```
Int[csc[e_.+f_.*x_]*(A_+B_.*csc[e_.+f_.*x_])/Sqrt[a_+b_.*csc[e_.+f_.*x_]],x_Symbol] :=
  (A-B)*Int[Csc[e+f*x]/Sqrt[a+b*Csc[e+f*x]],x] +
  B*Int[Csc[e+f*x]*(1+Csc[e+f*x])/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,e,f,A,B},x] && NeQ[a^2-b^2,0] && NeQ[A^2-B^2,0]
```

4: $\int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge A^2 - B^2 = 0 \wedge 2m \notin \mathbb{Z}$

Derivation: Integration by substitution

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge A^2 - B^2 = 0 \wedge 2m \notin \mathbb{Z}$, then

$$\int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \rightarrow$$

$$-\frac{2\sqrt{2} A (a+b \sec[e+fx])^m (A-B \sec[e+fx]) \sqrt{\frac{A+B \sec[e+fx]}{A}}}{B f \tan[e+fx] \left(\frac{A(a+b \sec[e+fx])}{aA+bB}\right)^m} \text{AppellF1}\left[\frac{1}{2}, -\frac{1}{2}, -m, \frac{3}{2}, \frac{A-B \sec[e+fx]}{2A}, \frac{b(A-B \sec[e+fx])}{Ab+aB}\right]$$

Program code:

```
Int[csc[e_.+f_.*x_]*(a_+b_.*csc[e_.+f_.*x_])^m*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  2*Sqrt[2]*A*(a+b*Csc[e+f*x])^m*(A-B*Csc[e+f*x])*Sqrt[(A+B*Csc[e+f*x])/A]/(B*f*Cot[e+f*x]*(A*(a+b*Csc[e+f*x])/(a*A+b*B))^m)*
  AppellF1[1/2, -(1/2), -m, 3/2, (A-B*Csc[e+f*x])/(2*A), (b*(A-B*Csc[e+f*x]))/(A*b+a*B)] /;
FreeQ[{a,b,A,B,e,f},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && EqQ[A^2-B^2,0] && Not[IntegerQ[2*m]]
```

$$\mathbf{5:} \int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0$$

Derivation: Algebraic expansion

$$\text{Basis: } A + Bz == \frac{Ab-aB}{b} + \frac{B}{b} (a + bz)$$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0$, then

$$\int \sec[e+fx] (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \rightarrow \frac{Ab-aB}{b} \int \sec[e+fx] (a+b \sec[e+fx])^m dx + \frac{B}{b} \int \sec[e+fx] (a+b \sec[e+fx])^{m+1} dx$$

Program code:

```
Int[csc[e_+f_*x_]*(a_+b_*csc[e_+f_*x_])^m_*(A_+B_*csc[e_+f_*x_]),x_Symbol] :=
  (A*b-a*B)/b*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m,x] + B/b*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1),x] /;
FreeQ[{a,b,A,B,e,f,m},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0]
```

3. $\int \sec[e+fx]^2 (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx$ when $Ab - aB \neq 0$

1: $\int \sec[e+fx]^2 (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx$ when $Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m < -\frac{1}{2}$

Derivation: ???

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m < -\frac{1}{2}$, then

$$\int \sec[e+fx]^2 (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \rightarrow$$

$$\frac{(Ab - aB) \tan[e+fx] (a+b \sec[e+fx])^m}{bf(2m+1)} +$$

$$\frac{1}{b^2(2m+1)} \int \sec[e+fx] (a+b \sec[e+fx])^{m+1} (m(Ab - aB) + bB(2m+1) \sec[e+fx]) dx$$

Program code:

```
Int[csc[e_.+f_.*x_]^2*(a_+b_.*csc[e_.+f_.*x_]^m*(A_+B_.*csc[e_.+f_.*x_] ),x_Symbol] :=
- (A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m/(b*f*(2*m+1)) +
1/(b^2*(2*m+1))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*Simp[A*b*m-a*B*m+b*B*(2*m+1)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && LtQ[m,-1/2]
```

2: $\int \sec[e+fx]^2 (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx$ when $Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m < -1$

Derivation: Nondegenerate secant recurrence 1a with $A \rightarrow aA$, $B \rightarrow Ab + aB$, $C \rightarrow bB$, $m \rightarrow 0$, $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m < -1$, then

$$\int \sec[e+fx]^2 (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \rightarrow$$

$$- \frac{a(Ab - aB) \tan[e+fx] (a+b \sec[e+fx])^{m+1}}{bf(m+1)(a^2 - b^2)} -$$

$$\frac{1}{b(m+1)(a^2-b^2)} \int \sec[e+fx] (a+b \sec[e+fx])^{m+1} (b(Ab-aB)(m+1) - (aAb(m+2) - B(a^2+b^2(m+1))) \sec[e+fx]) dx$$

Program code:

```
Int[csc[e_.+f_.**x_]^2*(a_+b_.**csc[e_.+f_.**x_])^m_*(A_+B_.**csc[e_.+f_.**x_]),x_Symbol] :=
  a*(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+1)*(a^2-b^2)) -
  1/(b*(m+1)*(a^2-b^2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*
    Simp[b*(A*b-a*B)*(m+1)-(a*A*b*(m+2)-B*(a^2+b^2*(m+1)))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[m,-1]
```

3: $\int \sec[e+fx]^2 (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx$ when $Ab-aB \neq 0 \wedge m \neq -1$

Derivation: Nondegenerate secant recurrence 1b with $A \rightarrow aA$, $B \rightarrow Ab+aB$, $C \rightarrow bB$, $m \rightarrow 0$, $p \rightarrow 0$

Rule: If $Ab-aB \neq 0 \wedge m \neq -1$, then

$$\int \sec[e+fx]^2 (a+b \sec[e+fx])^m (A+B \sec[e+fx]) dx \rightarrow$$

$$\frac{B \tan[e+fx] (a+b \sec[e+fx])^{m+1}}{b f (m+2)} + \frac{1}{b (m+2)} \int \sec[e+fx] (a+b \sec[e+fx])^m (b B (m+1) + (Ab(m+2) - aB) \sec[e+fx]) dx$$

Program code:

```
Int[csc[e_.+f_.**x_]^2*(a_+b_.**csc[e_.+f_.**x_])^m_*(A_+B_.**csc[e_.+f_.**x_]),x_Symbol] :=
  -B*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)/(b*f*(m+2)) +
  1/(b*(m+2))*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^m*Simp[b*B*(m+1)+(A*b*(m+2)-a*B)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,e,f,A,B,m},x] && NeQ[A*b-a*B,0] && Not[LtQ[m,-1]]
```

$$4. \int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \text{ when } A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$$

$$1. \int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \text{ when } A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m + n + 1 \neq 0$$

$$1: \int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \text{ when } A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m + n + 1 \neq 0 \wedge a A m - b B n \neq 0$$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m + n + 1 \neq 0 \wedge a A m - b B n \neq 0$, then

$$\int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \rightarrow - \frac{A \tan[e + f x] (a + b \sec[e + f x])^m (d \sec[e + f x])^n}{f n}$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*n) /;
FreeQ[{a,b,d,e,f,A,B,m,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && EqQ[m+n+1,0] && EqQ[a*A*m-b*B*n,0]
```

$$2. \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m+n+1 = 0 \wedge aAm - bBn \neq 0$$

$$1: \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m+n+1 = 0 \wedge m \leq -1$$

Derivation: Singly degenerate secant recurrence 2b with $m \rightarrow -n-2$, $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m+n+1 = 0 \wedge m \leq -1$, then

$$\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow \frac{(Ab - aB) \tan[e+fx] (a+b \sec[e+fx])^m (d \sec[e+fx])^n}{bf(2m+1)} + \frac{(aAm + bB(m+1))}{a^2(2m+1)} \int (a+b \sec[e+fx])^{m+1} (d \sec[e+fx])^n dx$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
- (A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(b*f*(2*m+1)) +
(a*A*m+b*B*(m+1))/(a^2*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && EqQ[m+n+1,0] && LeQ[m,-1]
```

$$\mathbf{2:} \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m+n+1 = 0 \wedge m \neq -1$$

Derivation: Singly degenerate secant recurrence 1c with $m \rightarrow -n-2$, $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m+n+1 = 0 \wedge m \neq -1$, then

$$\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow \\ - \frac{A \tan[e+fx] (a+b \sec[e+fx])^m (d \sec[e+fx])^n}{f n} - \frac{(aAm - bBn)}{b d n} \int (a+b \sec[e+fx])^m (d \sec[e+fx])^{n+1} dx$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*n) -
  (a*A*m-b*B*n)/(b*d*n)*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,A,B,m,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && EqQ[m+n+1,0] && Not[LeQ[m,-1]]
```

$$2. \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m \geq \frac{1}{2}$$

$$1. \int \sqrt{a+b \sec[e+fx]} (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 = 0$$

$$\mathbf{1:} \int \sqrt{a+b \sec[e+fx]} (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge Ab(2n+1) + 2aBn = 0$$

Derivation: Singly degenerate secant recurrence 1a with $B \rightarrow -\frac{Ab(3+2n)}{2a(1+n)}$, $m \rightarrow \frac{1}{2}$, $p \rightarrow 0$

Derivation: Singly degenerate secant recurrence 1b with $B \rightarrow -\frac{Ab(3+2n)}{2a(1+n)}$, $m \rightarrow \frac{1}{2}$, $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge Ab(2n+1) + 2aBn = 0$, then

$$\int \sqrt{a+b \sec[e+fx]} (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow \frac{2 b B \tan[e+fx] (d \sec[e+fx])^n}{f (2 n+1) \sqrt{a+b \sec[e+fx]}}$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_+f_.*x_]]*(d_.*csc[e_+f_.*x_])^n*(A_+B_.*csc[e_+f_.*x_]),x_Symbol] :=
  -2*b*B*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*(2*n+1)*Sqrt[a+b*Csc[e+f*x]]) /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && EqQ[A*b*(2*n+1)+2*a*B*n,0]
```

2. $\int \sqrt{a+b \sec[e+fx]} (d \sec[e+fx])^n (A+B \sec[e+fx]) dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 = 0 \wedge A b (2 n+1) + 2 a B n \neq 0$

1: $\int \sqrt{a+b \sec[e+fx]} (d \sec[e+fx])^n (A+B \sec[e+fx]) dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 = 0 \wedge A b (2 n+1) + 2 a B n \neq 0 \wedge n < 0$

Derivation: Singly degenerate secant recurrence 1a with $m \rightarrow \frac{1}{2}$, $p \rightarrow 0$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 = 0 \wedge A b (2 n+1) + 2 a B n \neq 0 \wedge n < 0$, then

$$\int \sqrt{a+b \sec[e+fx]} (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow$$

$$-\frac{A b^2 \tan[e+fx] (d \sec[e+fx])^n}{a f n \sqrt{a+b \sec[e+fx]}} + \frac{(A b (2 n+1) + 2 a B n)}{2 a d n} \int \sqrt{a+b \sec[e+fx]} (d \sec[e+fx])^{n+1} dx$$

Program code:

```
Int[Sqrt[a_+b_.*csc[e_+f_.*x_]]*(d_.*csc[e_+f_.*x_])^n*(A_+B_.*csc[e_+f_.*x_]),x_Symbol] :=
  A*b^2*Cot[e+f*x]*(d*Csc[e+f*x])^n/(a*f*n*Sqrt[a+b*Csc[e+f*x]]) +
  (A*b*(2*n+1)+2*a*B*n)/(2*a*d*n)*Int[Sqrt[a+b*Csc[e+f*x]]*(d*Csc[e+f*x])^(n+1),x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && NeQ[A*b*(2*n+1)+2*a*B*n,0] && LtQ[n,0]
```


$$2: \int \sqrt{a+b \sec[e+fx]} (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge Ab(2n+1) + 2aBn \neq 0 \wedge n \neq 0$$

Derivation: Singly degenerate secant recurrence 1b with $m \rightarrow \frac{1}{2}$, $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge Ab(2n+1) + 2aBn \neq 0 \wedge n \neq 0$, then

$$\int \sqrt{a+b \sec[e+fx]} (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow \frac{2bB \tan[e+fx] (d \sec[e+fx])^n}{f(2n+1) \sqrt{a+b \sec[e+fx]}} + \frac{Ab(2n+1) + 2aBn}{b(2n+1)} \int \sqrt{a+b \sec[e+fx]} (d \sec[e+fx])^n dx$$

Program code:

```
Int[Sqrt[a+b_*csc[e_+f_*x_]]*(d_*csc[e_+f_*x_])^n*(A+B_*csc[e_+f_*x_]),x_Symbol] :=
-2*b*B*Cot[e+f*x]*(d*Csc[e+f*x])^n/(f*(2*n+1)*Sqrt[a+b*Csc[e+f*x]]) +
(A*b*(2*n+1)+2*a*B*n)/(b*(2*n+1))*Int[Sqrt[a+b*Csc[e+f*x]]*(d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && NeQ[A*b*(2*n+1)+2*a*B*n,0] && Not[LtQ[n,0]]
```

$$2. \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m > \frac{1}{2}$$

$$1: \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m > \frac{1}{2} \wedge n < -1$$

Derivation: Singly degenerate secant recurrence 1a with $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m > \frac{1}{2} \wedge n < -1$, then

$$\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow -\frac{aA \tan[e+fx] (a+b \sec[e+fx])^{m-1} (d \sec[e+fx])^n}{fn}$$

$$\frac{b}{a d n} \int (a + b \sec[e + f x])^{m-1} (d \sec[e + f x])^{n+1} (a A (m - n - 1) - b B n - (a B n + A b (m + n)) \sec[e + f x]) dx$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a*A*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n/(f*n) -
  b/(a*d*n)*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^(n+1)*Simp[a*A*(m-n-1)-b*B*n-(a*B*n+A*b*(m+n))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && GtQ[m,1/2] && LtQ[n,-1]
```

$$2: \int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \text{ when } A b - a B \neq 0 \wedge a^2 - b^2 = 0 \wedge m > \frac{1}{2} \wedge n \neq -1$$

Derivation: Singly degenerate secant recurrence 1b with $p \rightarrow 0$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 = 0 \wedge m > \frac{1}{2} \wedge n \neq -1$, then

$$\int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \rightarrow$$

$$\frac{b B \tan[e + f x] (a + b \sec[e + f x])^{m-1} (d \sec[e + f x])^n}{f (m + n)} +$$

$$\frac{1}{d (m + n)} \int (a + b \sec[e + f x])^{m-1} (d \sec[e + f x])^n (a A d (m + n) + B (b d n) + (A b d (m + n) + a B d (2 m + n - 1)) \sec[e + f x]) dx$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -b*B*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n/(f*(m+n)) +
  1/(d*(m+n))*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n*Simp[a*A*d*(m+n)+B*(b*d*n)+(A*b*d*(m+n)+a*B*d*(2*m+n-1))*Csc[e+f*x],x],x]
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && GtQ[m,1/2] && Not[LtQ[n,-1]]
```

$$3. \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m < -\frac{1}{2}$$

$$1: \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m < -\frac{1}{2} \wedge n > 0$$

Derivation: Singly degenerate secant recurrence 2a with $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m < -\frac{1}{2} \wedge n > 0$, then

$$\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow$$

$$-\frac{d(Ab - aB) \tan[e+fx] (a+b \sec[e+fx])^m (d \sec[e+fx])^{n-1}}{af(2m+1)} -$$

$$\frac{1}{ab(2m+1)} \int (a+b \sec[e+fx])^{m+1} (d \sec[e+fx])^{n-1} (A(a d(n-1)) - B(b d(n-1)) - d(aB(m-n+1) + Ab(m+n)) \sec[e+fx]) dx$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
d*(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-1)/(a*f*(2*m+1)) -
1/(a*b*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)*
Simp[A*(a*d*(n-1))-B*(b*d*(n-1))-d*(a*B*(m-n+1)+A*b*(m+n))*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && LtQ[m,-1/2] && GtQ[n,0]
```

$$2: \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m < -\frac{1}{2} \wedge n \neq 0$$

Derivation: Singly degenerate secant recurrence 2b with $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge m < -\frac{1}{2} \wedge n \neq 0$, then

$$\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow$$

$$\frac{(Ab - aB) \tan[e+fx] (a+b \sec[e+fx])^m (d \sec[e+fx])^n}{bf(2m+1)} -$$

$$\frac{1}{a^2 (2m+1)} \int (a+b \sec[e+fx])^{m+1} (d \sec[e+fx])^n (bBn - aA(2m+n+1) + (Ab - aB)(m+n+1) \sec[e+fx]) dx$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(b*f*(2*m+1)) -
  1/(a^2*(2*m+1))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n*
    Simp[b*B*n-a*A*(2*m+n+1)+(A*b-a*B)*(m+n+1)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && LtQ[m,-1/2] && Not[GtQ[n,0]]
```

4: $\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx$ when $Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge n > 1$

Derivation: Singly degenerate secant recurrence 2c with $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 = 0 \wedge n > 1$, then

$$\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow$$

$$\frac{B d \tan[e+fx] (a+b \sec[e+fx])^m (d \sec[e+fx])^{n-1}}{f(m+n)} +$$

$$\frac{d}{b(m+n)} \int (a+b \sec[e+fx])^m (d \sec[e+fx])^{n-1} (bB(n-1) + (Ab(m+n) + aBm) \sec[e+fx]) dx$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  -B*d*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-1)/(f*(m+n)) +
  d/(b*(m+n))*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-1)*Simp[b*B*(n-1)+(A*b*(m+n)+a*B*m)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B,m},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && GtQ[n,1]
```

5: $\int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 = 0 \wedge n < 0$

Derivation: Singly degenerate secant recurrence 1c with $p \rightarrow 0$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 = 0 \wedge n < 0$, then

$$\int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \rightarrow$$

$$- \frac{A \tan[e + f x] (a + b \sec[e + f x])^m (d \sec[e + f x])^n}{f n} -$$

$$\frac{1}{b d n} \int (a + b \sec[e + f x])^m (d \sec[e + f x])^{n+1} (a A m - b B n - A b (m + n + 1) \sec[e + f x]) dx$$

Program code:

```
Int[(a_+b_.*csc[e_+f_.*x_])^m_*(d_.*csc[e_+f_.*x_])^n_*(A_+B_.*csc[e_+f_.*x_]),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*n) -
  1/(b*d*n)*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n+1)*Simp[a*A*m-b*B*n-A*b*(m+n+1)*Csc[e+f*x],x],x] /;
FreeQ[{a,b,d,e,f,A,B,m},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0] && LtQ[n,0]
```

6: $\int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 = 0$

Derivation: Algebraic expansion

Baisi: $A + B z = \frac{A b - a B}{b} + \frac{B (a + b z)}{b}$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 = 0$, then

$$\int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \rightarrow \frac{A b - a B}{b} \int (a + b \sec[e + f x])^m (d \sec[e + f x])^n dx + \frac{B}{b} \int (a + b \sec[e + f x])^{m+1} (d \sec[e + f x])^n dx$$

Program code:

```
Int[(a_+b_.*csc[e_+f_.*x_])^m_*(d_.*csc[e_+f_.*x_])^n_*(A_+B_.*csc[e_+f_.*x_]),x_Symbol] :=
  (A*b-a*B)/b*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n,x] +
  B/b*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n,x] /;
FreeQ[{a,b,d,e,f,A,B,m},x] && NeQ[A*b-a*B,0] && EqQ[a^2-b^2,0]
```

5. $\int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$

1. $\int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m > 1$

1: $\int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m > 1 \wedge n \leq -1$

Derivation: Nondegenerate secant recurrence 1a with $A \rightarrow a A$, $B \rightarrow A b + a B$, $C \rightarrow b B$, $m \rightarrow m - 1$, $p \rightarrow 0$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m > 1 \wedge n \leq -1$, then

$$\int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \rightarrow$$

$$- \frac{a A \tan[e+fx] (a+b \sec[e+fx])^{m-1} (d \sec[e+fx])^n}{f n} + \frac{1}{d n} \int (a+b \sec[e+fx])^{m-2} (d \sec[e+fx])^{n+1} .$$

$$(a (a B n - A b (m-n-1)) + (2 a b B n + A (b^2 n + a^2 (1+n))) \sec[e+fx] + b (b B n + a A (m+n)) \sec[e+fx]^2) dx$$

Program code:

```
Int[(a_+b_.*csc[e_+f_.*x_])^m_*(d_.*csc[e_+f_.*x_])^n_*(A_+B_.*csc[e_+f_.*x_]),x_Symbol] :=
  a*A*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n/(f*n) +
  1/(d*n)*Int[(a+b*Csc[e+f*x])^(m-2)*(d*Csc[e+f*x])^(n+1)*
    Simp[a*(a*B*n-A*b*(m-n-1))+(2*a*b*B*n+A*(b^2*n+a^2*(1+n)))*Csc[e+f*x]+b*(b*B*n+a*A*(m+n))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && GtQ[m,1] && LeQ[n,-1]
```

$$\mathbf{2:} \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m > 1 \wedge n \neq -1$$

Derivation: Nondegenerate secant recurrence 1b with $A \rightarrow aA$, $B \rightarrow Ab + aB$, $C \rightarrow bB$, $m \rightarrow m-1$, $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m > 1 \wedge n \neq -1$, then

$$\begin{aligned} & \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow \\ & \frac{bB \tan[e+fx] (a+b \sec[e+fx])^{m-1} (d \sec[e+fx])^n}{f(m+n)} + \\ & \frac{1}{m+n} \int (a+b \sec[e+fx])^{m-2} (d \sec[e+fx])^n \cdot \\ & (a^2 A(m+n) + a b B n + (a(2Ab + aB)(m+n) + b^2 B(m+n-1)) \sec[e+fx] + b(Ab(m+n) + aB(2m+n-1)) \sec[e+fx]^2) dx \end{aligned}$$

Program code:

```
Int[(a_+b_.*csc[e_+f_.*x_])^m_*(d_.*csc[e_+f_.*x_])^n_*(A_+B_.*csc[e_+f_.*x_]),x_Symbol] :=
-b*B*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^n/(f*(m+n)) +
1/(m+n)*Int[(a+b*Csc[e+f*x])^(m-2)*(d*Csc[e+f*x])^n*
Simp[a^2*A*(m+n)+a*b*B*n+(a*(2*A*b+a*B)*(m+n)+b^2*B*(m+n-1))*Csc[e+f*x]+b*(A*b*(m+n)+a*B*(2*m+n-1))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && GtQ[m,1] && Not[IGtQ[n,1] && Not[IntegerQ[m]]]
```

$$\mathbf{2.} \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m < -1$$

$$\mathbf{1.} \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m < -1 \wedge n > 0$$

$$\mathbf{1:} \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m < -1 \wedge 0 < n < 1$$

Derivation: Nondegenerate secant recurrence 1a with $C \rightarrow 0$, $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m < -1 \wedge 0 < n < 1$, then

$$\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow$$

$$\frac{d (A b - a B) \tan[e+fx] (a+b \sec[e+fx])^{m+1} (d \sec[e+fx])^{n-1}}{f (m+1) (a^2 - b^2)} +$$

$$\frac{1}{(m+1) (a^2 - b^2)} \int (a+b \sec[e+fx])^{m+1} (d \sec[e+fx])^{n-1} \cdot$$

$$(d (n-1) (A b - a B) + d (a A - b B) (m+1) \sec[e+fx] - d (A b - a B) (m+n+1) \sec[e+fx]^2) dx$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*Csc[e_.+f_.*x_]),x_Symbol] :=
  -d*(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)/(f*(m+1)*(a^2-b^2)) +
  1/((m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-1)*
    Simp[d*(n-1)*(A*b-a*B)+d*(a*A-b*B)*(m+1)*Csc[e+f*x]-d*(A*b-a*B)*(m+n+1)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[m,-1] && LtQ[0,n,1]
```

$$2: \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m < -1 \wedge n > 1$$

Derivation: Nondegenerate secant recurrence 1a with $A \rightarrow aA$, $B \rightarrow Ab + aB$, $C \rightarrow bB$, $m \rightarrow m-1$, $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m < -1 \wedge n > 1$, then

$$\begin{aligned} & \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow \\ & - \left((a d^2 (Ab - aB) \tan[e+fx] (a+b \sec[e+fx])^{m+1} (d \sec[e+fx])^{n-2}) / (b f (m+1) (a^2 - b^2)) \right) - \\ & \quad \frac{d}{b (m+1) (a^2 - b^2)} \int (a+b \sec[e+fx])^{m+1} (d \sec[e+fx])^{n-2} \cdot \\ & (a d (Ab - aB) (n-2) + b d (Ab - aB) (m+1) \sec[e+fx] - (aAbd(m+n) - dB(a^2(n-1) + b^2(m+1))) \sec[e+fx]^2) dx \end{aligned}$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  a*d^2*(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-2)/(b*f*(m+1)*(a^2-b^2)) -
  d/(b*(m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-2)*
  Simp[a*d*(A*b-a*B)*(n-2)+b*d*(A*b-a*B)*(m+1)*Csc[e+f*x]-(a*A*b*d*(m+n)-d*B*(a^2*(n-1)+b^2*(m+1)))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[m,-1] && GtQ[n,1]
```

$$2: \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \text{ when } Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m < -1 \wedge n \neq 0$$

Derivation: Nondegenerate secant recurrence 1c with $C \rightarrow 0$, $p \rightarrow 0$

Rule: If $Ab - aB \neq 0 \wedge a^2 - b^2 \neq 0 \wedge m < -1 \wedge n \neq 0$, then

$$\begin{aligned} & \int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow \\ & - \frac{b (Ab - aB) \tan[e+fx] (a+b \sec[e+fx])^{m+1} (d \sec[e+fx])^n}{a f (m+1) (a^2 - b^2)} + \end{aligned}$$

$$\frac{1}{a(m+1)(a^2-b^2)} \int (a+b \sec[e+fx])^{m+1} (d \sec[e+fx])^n \cdot \\ (A(a^2(m+1)-b^2(m+n+1)) + abBn - a(Ab-aB)(m+1) \sec[e+fx] + b(Ab-aB)(m+n+2) \sec[e+fx]^2) dx$$

Program code:

```
Int[(a_+b_.*csc[e_+f_.*x_])^m_*(d_.*csc[e_+f_.*x_])^n_*(A_+B_.*csc[e_+f_.*x_]),x_Symbol] :=
  b*(A*b-a*B)*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n/(a*f*(m+1)*(a^2-b^2)) +
  1/(a*(m+1)*(a^2-b^2))*Int[(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n*
    Simp[A*(a^2*(m+1)-b^2*(m+n+1))+a*b*B*n-a*(A*b-a*B)*(m+1)*Csc[e+f*x]+b*(A*b-a*B)*(m+n+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[m,-1] && Not[ILtQ[m+1/2,0] && ILtQ[n,0]]
```

$$3. \int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \text{ when } A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge 0 < m < 1$$

$$1: \int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \text{ when } A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge 0 < m < 1 \wedge n > 0$$

Derivation: Nondegenerate secant recurrence 1b with $A \rightarrow A c$, $B \rightarrow B c + A d$, $C \rightarrow B d$, $n \rightarrow n - 1$, $p \rightarrow 0$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge 0 < m < 1 \wedge n > 0$, then

$$\begin{aligned} & \int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \rightarrow \\ & \frac{B d \tan[e + f x] (a + b \sec[e + f x])^m (d \sec[e + f x])^{n-1}}{f (m+n)} + \\ & \frac{d}{m+n} \int (a + b \sec[e + f x])^{m-1} (d \sec[e + f x])^{n-1} \cdot \\ & (a B (n-1) + (b B (m+n-1) + a A (m+n)) \sec[e + f x] + (a B m + A b (m+n)) \sec[e + f x]^2) dx \end{aligned}$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
-B*d*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-1)/(f*(m+n)) +
d/(m+n)*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^(n-1)*
Simp[a*B*(n-1)+(b*B*(m+n-1)+a*A*(m+n))*Csc[e+f*x]+(a*B*m+A*b*(m+n))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[0,m,1] && GtQ[n,0]
```

$$2: \int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \text{ when } A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge 0 < m < 1 \wedge n \leq -1$$

Derivation: Nondegenerate secant recurrence 1a with $C \rightarrow 0$, $p \rightarrow 0$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge 0 < m < 1 \wedge n \leq -1$, then

$$\int (a + b \sec[e + f x])^m (d \sec[e + f x])^n (A + B \sec[e + f x]) dx \rightarrow$$

$$- \frac{A \tan[e+fx] (a+b \sec[e+fx])^m (d \sec[e+fx])^n}{f n} - \frac{1}{d n} \int (a+b \sec[e+fx])^{m-1} (d \sec[e+fx])^{n+1} \cdot (A b m - a B n - (b B n + a A (n+1)) \sec[e+fx] - A b (m+n+1) \sec[e+fx]^2) dx$$

Program code:

```
Int[(a_+b_.*csc[e_+f_.*x_])^m_*(d_.*csc[e_+f_.*x_])^n_*(A_+B_.*csc[e_+f_.*x_]),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n/(f*n) -
  1/(d*n)*Int[(a+b*Csc[e+f*x])^(m-1)*(d*Csc[e+f*x])^(n+1)*
    Simp[A*b*m-a*B*n-(b*B*n+a*A*(n+1))*Csc[e+f*x]-A*b*(m+n+1)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LtQ[0,m,1] && LeQ[n,-1]
```

4: $\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge n > 1 \wedge m+n \neq 0$

Derivation: Nondegenerate secant recurrence 1b with $A \rightarrow a A$, $B \rightarrow A b + a B$, $C \rightarrow b B$, $m \rightarrow m-1$, $p \rightarrow 0$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge n > 1 \wedge m+n \neq 0$, then

$$\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow \frac{B d^2 \tan[e+fx] (a+b \sec[e+fx])^{m+1} (d \sec[e+fx])^{n-2}}{b f (m+n)} + \frac{d^2}{b (m+n)} \int (a+b \sec[e+fx])^m (d \sec[e+fx])^{n-2} (a B (n-2) + B b (m+n-1) \sec[e+fx] + (A b (m+n) - a B (n-1)) \sec[e+fx]^2) dx$$

Program code:

```
Int[(a_+b_.*csc[e_+f_.*x_])^m_*(d_.*csc[e_+f_.*x_])^n_*(A_+B_.*csc[e_+f_.*x_]),x_Symbol] :=
  -B*d^2*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^(n-2)/(b*f*(m+n)) +
  d^2/(b*(m+n))*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n-2)*
    Simp[a*B*(n-2)+B*b*(m+n-1)*Csc[e+f*x]+(A*b*(m+n)-a*B*(n-1))*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,m},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && GtQ[n,1] && NeQ[m+n,0] && Not[IGtQ[m,1]]
```

5: $\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge n \leq -1$

Derivation: Nondegenerate secant recurrence 1c with $C \rightarrow 0$, $p \rightarrow 0$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0 \wedge n \leq -1$, then

$$\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow$$

$$-\frac{A \tan[e+fx] (a+b \sec[e+fx])^{m+1} (d \sec[e+fx])^n}{a f n} +$$

$$\frac{1}{a d n} \int (a+b \sec[e+fx])^m (d \sec[e+fx])^{n+1} (a B n - A b (m+n+1) + A a (n+1) \sec[e+fx] + A b (m+n+2) \sec[e+fx]^2) dx$$

Program code:

```
Int[(a+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A*Cot[e+f*x]*(a+b*Csc[e+f*x])^(m+1)*(d*Csc[e+f*x])^n/(a*f*n) +
  1/(a*d*n)*Int[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^(n+1)*
    Simp[a*B*n-A*b*(m+n+1)+A*a*(n+1)*Csc[e+f*x]+A*b*(m+n+2)*Csc[e+f*x]^2,x],x] /;
FreeQ[{a,b,d,e,f,A,B,m},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0] && LeQ[n,-1]
```

6: $\int \frac{A+B \sec[e+fx]}{\sqrt{d \sec[e+fx]} \sqrt{a+b \sec[e+fx]}} dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+B z}{\sqrt{d z} \sqrt{a+b z}} == \frac{A \sqrt{a+b z}}{a \sqrt{d z}} - \frac{(A b - a B) \sqrt{d z}}{a d \sqrt{a+b z}}$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$, then

$$\int \frac{A + B \sec[e + f x]}{\sqrt{d \sec[e + f x]} \sqrt{a + b \sec[e + f x]}} dx \rightarrow \frac{A}{a} \int \frac{\sqrt{a + b \sec[e + f x]}}{\sqrt{d \sec[e + f x]}} dx - \frac{A b - a B}{a d} \int \frac{\sqrt{d \sec[e + f x]}}{\sqrt{a + b \sec[e + f x]}} dx$$

Program code:

```
Int[(A_+B_.*csc[e_+f_.*x_])/(Sqrt[d_.*csc[e_+f_.*x_]]*Sqrt[a_+b_.*csc[e_+f_.*x_]]),x_Symbol] :=
  A/a*Int[Sqrt[a+b*Csc[e+f*x]]/Sqrt[d*Csc[e+f*x]],x] -
  (A*b-a*B)/(a*d)*Int[Sqrt[d*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0]
```

7: $\int \frac{\sqrt{d \sec[e + f x]} (A + B \sec[e + f x])}{\sqrt{a + b \sec[e + f x]}} dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$

Derivation: Algebraic expansion

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$, then

$$\int \frac{\sqrt{d \sec[e + f x]} (A + B \sec[e + f x])}{\sqrt{a + b \sec[e + f x]}} dx \rightarrow A \int \frac{\sqrt{d \sec[e + f x]}}{\sqrt{a + b \sec[e + f x]}} dx + \frac{B}{d} \int \frac{(d \sec[e + f x])^{3/2}}{\sqrt{a + b \sec[e + f x]}} dx$$

Program code:

```
Int[Sqrt[d_.*csc[e_+f_.*x_]]*(A_+B_.*csc[e_+f_.*x_])/Sqrt[a_+b_.*csc[e_+f_.*x_]],x_Symbol] :=
  A*Int[Sqrt[d*Csc[e+f*x]]/Sqrt[a+b*Csc[e+f*x]],x] +
  B/d*Int[(d*Csc[e+f*x])^(3/2)/Sqrt[a+b*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0]
```

8:
$$\int \frac{\sqrt{a+b \sec[e+fx]} (A+B \sec[e+fx])}{\sqrt{d \sec[e+fx]}} dx \text{ when } A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$$

Derivation: Algebraic expansion

Basis:
$$\frac{A+Bz}{\sqrt{dz}} == \frac{B\sqrt{dz}}{d} + \frac{A}{\sqrt{dz}}$$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$, then

$$\int \frac{\sqrt{a+b \sec[e+fx]} (A+B \sec[e+fx])}{\sqrt{d \sec[e+fx]}} dx \rightarrow \frac{B}{d} \int \sqrt{a+b \sec[e+fx]} \sqrt{d \sec[e+fx]} dx + A \int \frac{\sqrt{a+b \sec[e+fx]}}{\sqrt{d \sec[e+fx]}} dx$$

Program code:

```
Int[Sqrt[a+b_.*csc[e_.+f_.*x_]]*(A+B_.*csc[e_.+f_.*x_])/Sqrt[d_.*csc[e_.+f_.*x_]],x_Symbol] :=
  B/d*Int[Sqrt[a+b*Csc[e+f*x]]*Sqrt[d*Csc[e+f*x]],x] +
  A*Int[Sqrt[a+b*Csc[e+f*x]]/Sqrt[d*Csc[e+f*x]],x] /;
FreeQ[{a,b,d,e,f,A,B},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0]
```


9: $\int \frac{(d \sec[e+fx])^n (A+B \sec[e+fx])}{a+b \sec[e+fx]} dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{A+Bz}{a+bz} == \frac{A}{a} - \frac{(Ab-aB)(dz)}{ad(a+bz)}$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$, then

$$\int \frac{(d \sec[e+fx])^n (A+B \sec[e+fx])}{a+b \sec[e+fx]} dx \rightarrow \frac{A}{a} \int (d \sec[e+fx])^n dx - \frac{Ab-aB}{ad} \int \frac{(d \sec[e+fx])^{n+1}}{a+b \sec[e+fx]} dx$$

Program code:

```
Int[(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_])/(a_+b_.*csc[e_.+f_.*x_]),x_Symbol] :=
  A/a*Int[(d*Csc[e+f*x])^n,x] - (A*b-a*B)/(a*d)*Int[(d*Csc[e+f*x])^(n+1)/(a+b*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,A,B,n},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0]
```

X: $\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx$ when $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$

Rule: If $A b - a B \neq 0 \wedge a^2 - b^2 \neq 0$, then

$$\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx \rightarrow$$

$$\int (a+b \sec[e+fx])^m (d \sec[e+fx])^n (A+B \sec[e+fx]) dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_*(d_.*csc[e_.+f_.*x_])^n_*(A_+B_.*csc[e_.+f_.*x_]),x_Symbol] :=
  Unintegrable[(a+b*Csc[e+f*x])^m*(d*Csc[e+f*x])^n*(A+B*Csc[e+f*x]),x] /;
FreeQ[{a,b,d,e,f,A,B,m,n},x] && NeQ[A*b-a*B,0] && NeQ[a^2-b^2,0]
```

Rules for integrands of the form $(a + b \sec[e + f x])^m (c + d \sec[e + f x])^n (A + B \sec[e + f x])^p$

1. $\int (a + b \sec[e + f x])^m (c + d \sec[e + f x])^n (A + B \sec[e + f x])^p dx$ when $b c + a d = 0 \wedge a^2 - b^2 = 0$

x: $\int (a + b \sec[e + f x])^m (c + d \sec[e + f x])^n (A + B \sec[e + f x])^p dx$ when $b c + a d = 0 \wedge a^2 - b^2 = 0 \wedge m \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b c + a d = 0 \wedge a^2 - b^2 = 0$, then $(a + b \sec[z]) (c + d \sec[z]) = -a c \tan[z]^2$

Rule: If $b c + a d = 0 \wedge a^2 - b^2 = 0 \wedge m \in \mathbb{Z}$, then

$$\int (a + b \sec[e + f x])^m (c + d \sec[e + f x])^n (A + B \sec[e + f x])^p dx \rightarrow (-a c)^m \int \tan[e + f x]^{2m} (c + d \sec[e + f x])^{n-m} (A + B \sec[e + f x])^p dx$$

Program code:

```
(* Int[(a_+b_.*csc[e_+f_.*x_])^m_.*(c_+d_.*csc[e_+f_.*x_])^n_.*(A_+B_.*csc[e_+f_.*x_])^p_.,x_Symbol] :=
  (-a*c)^m*Int[Cot[e+f*x]^(2*m)*(c+d*Csc[e+f*x])^(n-m)*(A+B*Csc[e+f*x])^p,x] /;
FreeQ[{a,b,c,d,e,f,A,B,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegerQ[m] &&
Not[IntegerQ[n] && (LtQ[m,0] && GtQ[n,0] || LtQ[0,n,m] || LtQ[m,n,0])] *)
```

1: $\int (a + b \sec[e + f x])^m (c + d \sec[e + f x])^n (A + B \sec[e + f x])^p dx$ when $b c + a d = 0 \wedge a^2 - b^2 = 0 \wedge (m | n | p) \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $b c + a d = 0 \wedge a^2 - b^2 = 0$, then $(a + b \sec[z]) (c + d \sec[z]) = -a c \tan[z]^2$

Rule: If $b c + a d = 0 \wedge a^2 - b^2 = 0 \wedge (m | n | p) \in \mathbb{Z}$, then

$$\int (a + b \sec[e + f x])^m (c + d \sec[e + f x])^n (A + B \sec[e + f x])^p dx \rightarrow (-a c)^m \int \tan[e + f x]^{2m} (c + d \sec[e + f x])^{n-m} (A + B \sec[e + f x])^p dx$$

$$\rightarrow (-a c)^m \int \frac{\sin[e+fx]^{2m} (d+c \cos[e+fx])^{n-m} (B+A \cos[e+fx])^p}{\cos[e+fx]^{m+n+p}} dx$$

Program code:

```
Int[(a_+b_.*csc[e_+f_.*x_])^m_.*(c_+d_.*csc[e_+f_.*x_])^n_.*(A_+B_.*csc[e_+f_.*x_])^p_,x_Symbol] :=
  (-a*c)^m*Int[Cos[e+f*x]^(2*m)*(d+c*Sin[e+f*x])^(n-m)*(B+A*Sin[e+f*x])^p/Sin[e+f*x]^(m+n+p),x] /;
FreeQ[{a,b,c,d,e,f,A,B,n,p},x] && EqQ[b*c+a*d,0] && EqQ[a^2-b^2,0] && IntegersQ[m,n,p]
```