Rules for integrands involving product logarithm functions

1. $\int u\,(c\,\mathrm{ProductLog}[a+b\,x])^p\,dx$

   1. $\int (c\,\mathrm{ProductLog}[a+b\,x])^p\,dx$

      1: $\int (c\,\mathrm{ProductLog}[a+b\,x])^p\,dx$ when $p<-1$

Rule: If $p<-1$, then

$$\int (c\,\mathrm{ProductLog}[a+b\,x])^p\,dx \longrightarrow \frac{(a+b\,x)\,(c\,\mathrm{ProductLog}[a+b\,x])^p}{b\,(p+1)}+\frac{p}{c\,(p+1)}\int \frac{(c\,\mathrm{ProductLog}[a+b\,x])^{p+1}}{1+\mathrm{ProductLog}[a+b\,x]}\,dx$$

Program code:

```
Int[(c_.*ProductLog[a_.+b_.*x_])^p_,x_Symbol] :=
  (a+b*x)*(c*ProductLog[a+b*x])^p/(b*(p+1)) +
  p/(c*(p+1))*Int[(c*ProductLog[a+b*x])^(p+1)/(1+ProductLog[a+b*x]),x] /;
FreeQ[{a,b,c},x] && LtQ[p,-1]
```

2: $\int (c\,\mathrm{ProductLog}[a+b\,x])^p\,dx$ when $p \nless -1$

Derivation: Integration by parts

Rule: If $p \nless -1$, then

$$\int (c\,\mathrm{ProductLog}[a+b\,x])^p\,dx \rightarrow \frac{(a+b\,x)\,(c\,\mathrm{ProductLog}[a+b\,x])^p}{b} - p\int \frac{(c\,\mathrm{ProductLog}[a+b\,x])^p}{1+\mathrm{ProductLog}[a+b\,x]}\,dx$$

Program code:

```
Int[(c_.*ProductLog[a_.+b_.*x_])^p_.,x_Symbol] :=
  (a+b*x)*(c*ProductLog[a+b*x])^p/b -
  p*Int[(c*ProductLog[a+b*x])^p/(1+ProductLog[a+b*x]),x] /;
FreeQ[{a,b,c},x] && Not[LtQ[p,-1]]
```

2: $\int (e+f\,x)^m\,(c\,\mathrm{ProductLog}[a+b\,x])^p\,dx$ when $m \in \mathbb{Z}^+$

Derivation: Integration by substitution

Rule: If $m \in \mathbb{Z}^+$, then

$$\int (e+f\,x)^m\,(c\,\mathrm{ProductLog}[a+b\,x])^p\,dx \rightarrow \frac{1}{b^{m+1}}\,\mathrm{Subst}\Big[\int (c\,\mathrm{ProductLog}[x])^p\,\mathrm{ExpandIntegrand}\big[(b\,e-a\,f+f\,x)^m,\,x\big]\,dx,\,x,\,a+b\,x\Big]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(c_.*ProductLog[a_+b_.*x_])^p_.,x_Symbol] :=
  1/b^(m+1)*Subst[Int[ExpandIntegrand[(c*ProductLog[x])^p,(b*e-a*f+f*x)^m,x],x],x,a+b*x] /;
FreeQ[{a,b,c,e,f,p},x] && IGtQ[m,0]
```

2. $\int u\,(c\,\text{ProductLog}[a\,x^n])^p\,dx$

   1. $\int (c\,\text{ProductLog}[a\,x^n])^p\,dx$

      1: $\int (c\,\text{ProductLog}[a\,x^n])^p\,dx$ when $n\,(p-1) == -1\ \lor\ \left(p-\frac{1}{2}\in\mathbb{Z}\ \land\ n\left(p-\frac{1}{2}\right) == -1\right)$

Derivation: Integration by parts

Rule: If $n\,(p-1) == -1\ \lor\ \left(p-\frac{1}{2}\in\mathbb{Z}\ \land\ n\left(p-\frac{1}{2}\right) == -1\right)$, then

$$\int (c\,\text{ProductLog}[a\,x^n])^p\,dx \longrightarrow x\,(c\,\text{ProductLog}[a\,x^n])^p - n\,p\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{1+\text{ProductLog}[a\,x^n]}\,dx$$

Program code:

```
Int[(c_.*ProductLog[a_.*x_^n_])^p_.,x_Symbol] :=
  x*(c*ProductLog[a*x^n])^p -
  n*p*Int[(c*ProductLog[a*x^n])^p/(1+ProductLog[a*x^n]),x] /;
FreeQ[{a,c,n,p},x] && (EqQ[n*(p-1),-1] || IntegerQ[p-1/2] && EqQ[n*(p-1/2),-1])
```

2: $\int (c\,\text{ProductLog}[a\,x^n])^p\,dx$ when $(p \in \mathbb{Z} \land n\,(p+1) == -1) \lor \left(p - \frac{1}{2} \in \mathbb{Z} \land n\left(p + \frac{1}{2}\right) == -1\right)$

Rule: If $(p \in \mathbb{Z} \land n\,(p+1) == -1) \lor \left(p - \frac{1}{2} \in \mathbb{Z} \land n\left(p + \frac{1}{2}\right) == -1\right)$, then

$$\int (c\,\text{ProductLog}[a\,x^n])^p\,dx \to \frac{x\,(c\,\text{ProductLog}[a\,x^n])^p}{n\,p+1} + \frac{n\,p}{c\,(n\,p+1)} \int \frac{(c\,\text{ProductLog}[a\,x^n])^{p+1}}{1+\text{ProductLog}[a\,x^n]}\,dx$$

Program code:

```
Int[(c_.*ProductLog[a_.*x_^n_])^p_.,x_Symbol] :=
  x*(c*ProductLog[a*x^n])^p/(n*p+1) +
  n*p/(c*(n*p+1))*Int[(c*ProductLog[a*x^n])^(p+1)/(1+ProductLog[a*x^n]),x] /;
FreeQ[{a,c,n},x] && (IntegerQ[p] && EqQ[n*(p+1),-1] || IntegerQ[p-1/2] && EqQ[n*(p+1/2),-1])
```

3: $\int (c\,\text{ProductLog}[a\,x^n])^p\,dx$ when $n \in \mathbb{Z}^-$

Derivation: Integration by substitution

Basis: $\int f[x]\,dx == -\text{Subst}\left[\int \frac{f\left[\frac{1}{x}\right]}{x^2}\,dx,\ x,\ \frac{1}{x}\right]$

Rule: If $n \in \mathbb{Z}^-$, then

$$\int (c\,\text{ProductLog}[a\,x^n])^p\,dx \to -\text{Subst}\left[\int \frac{(c\,\text{ProductLog}[a\,x^{-n}])^p}{x^2}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[(c_.*ProductLog[a_.*x_^n_])^p_.,x_Symbol] :=
  -Subst[Int[(c*ProductLog[a*x^(-n)])^p/x^2,x],x,1/x] /;
FreeQ[{a,c,p},x] && ILtQ[n,0]
```

2. $\int x^m\,(c\,\text{ProductLog}[a\,x^n])^p\,dx$

   1: $\int x^m\,(c\,\text{ProductLog}[a\,x^n])^p\,dx$ when $m \neq -1 \land \left(p-\frac{1}{2}\in\mathbb{Z} \land 2\left(p+\frac{m+1}{n}\right)\in\mathbb{Z}^+ \lor p-\frac{1}{2}\notin\mathbb{Z} \land p+\frac{m+1}{n}+1\in\mathbb{Z}^+\right)$

Derivation: Integration by parts

Rule: If $m \neq -1 \land \left(p-\frac{1}{2}\in\mathbb{Z} \land 2\left(p+\frac{m+1}{n}\right)\in\mathbb{Z}^+ \lor p-\frac{1}{2}\notin\mathbb{Z} \land p+\frac{m+1}{n}+1\in\mathbb{Z}^+\right)$, then

$$\int x^m\,(c\,\text{ProductLog}[a\,x^n])^p\,dx \longrightarrow \frac{x^{m+1}\,(c\,\text{ProductLog}[a\,x^n])^p}{m+1}-\frac{n\,p}{m+1}\int\frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{1+\text{ProductLog}[a\,x^n]}\,dx$$

Program code:

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_.,x_Symbol] :=
  x^(m+1)*(c*ProductLog[a*x^n])^p/(m+1) -
  n*p/(m+1)*Int[x^m*(c*ProductLog[a*x^n])^p/(1+ProductLog[a*x^n]),x] /;
FreeQ[{a,c,m,n,p},x] && NeQ[m,-1] &&
(IntegerQ[p-1/2] && IGtQ[2*Simplify[p+(m+1)/n],0] || Not[IntegerQ[p-1/2]] && IGtQ[Simplify[p+(m+1)/n]+1,0])
```

2: $\int x^m\,(c\,\text{ProductLog}[a\,x^n])^p\,dx$ when $m == -1 \lor \left(p-\frac{1}{2}\in\mathbb{Z} \land p+\frac{m+1}{n}-\frac{1}{2}\in\mathbb{Z}^-\right) \lor \left(p-\frac{1}{2}\notin\mathbb{Z} \land p+\frac{m+1}{n}\in\mathbb{Z}^-\right)$

Rule: If $m == -1 \lor \left(p-\frac{1}{2}\in\mathbb{Z} \land p+\frac{m+1}{n}-\frac{1}{2}\in\mathbb{Z}^-\right) \lor \left(p-\frac{1}{2}\notin\mathbb{Z} \land p+\frac{m+1}{n}\in\mathbb{Z}^-\right)$, then

$$\int x^m\,(c\,\text{ProductLog}[a\,x^n])^p\,dx \longrightarrow \frac{x^{m+1}\,(c\,\text{ProductLog}[a\,x^n])^p}{m+n\,p+1}+\frac{n\,p}{c\,(m+n\,p+1)}\int\frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^{p+1}}{1+\text{ProductLog}[a\,x^n]}\,dx$$

Program code:

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_.,x_Symbol] :=
  x^(m+1)*(c*ProductLog[a*x^n])^p/(m+n*p+1) +
  n*p/(c*(m+n*p+1))*Int[x^m*(c*ProductLog[a*x^n])^(p+1)/(1+ProductLog[a*x^n]),x] /;
FreeQ[{a,c,m,n,p},x] &&
(EqQ[m,-1] || IntegerQ[p-1/2] && ILtQ[Simplify[p+(m+1)/n]-1/2,0] || Not[IntegerQ[p-1/2]] && ILtQ[Simplify[p+(m+1)/n],0])
```

3: $\int x^m\,(c\,\text{ProductLog}[a\,x])^p\,dx$

Derivation: Algebraic simplification

Basis: $1 = \frac{1}{1+z} + \frac{z}{1+z}$

Rule:

$$\int x^m\,(c\,\text{ProductLog}[a\,x])^p\,dx \to \int \frac{x^m\,(c\,\text{ProductLog}[a\,x])^p}{1+\text{ProductLog}[a\,x]}\,dx + \frac{1}{c}\int \frac{x^m\,(c\,\text{ProductLog}[a\,x])^{p+1}}{1+\text{ProductLog}[a\,x]}\,dx$$

Program code:

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_])^p_.,x_Symbol] :=
  Int[x^m*(c*ProductLog[a*x])^p/(1+ProductLog[a*x]),x] +
  1/c*Int[x^m*(c*ProductLog[a*x])^(p+1)/(1+ProductLog[a*x]),x] /;
FreeQ[{a,c,m},x]
```

4: $\int x^m\,(c\,\text{ProductLog}[a\,x^n])^p\,dx$ when $n\in\mathbb{Z}^-\land m\in\mathbb{Z}\land m\neq -1$

Derivation: Integration by substitution

Basis: $\int f[x]\,dx = -\text{Subst}\left[\int \frac{f\left[\frac{1}{x}\right]}{x^2}\,dx,\ x,\ \frac{1}{x}\right]$

Rule: If $n\in\mathbb{Z}^-\land m\in\mathbb{Z}\land m\neq -1$, then

$$\int x^m\,(c\,\text{ProductLog}[a\,x^n])^p\,dx \longrightarrow -\text{Subst}\left[\int \frac{(c\,\text{ProductLog}[a\,x^{-n}])^p}{x^{m+2}}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_])^p_.,x_Symbol] :=
  -Subst[Int[(c*ProductLog[a*x^(-n)])^p/x^(m+2),x],x,1/x] /;
FreeQ[{a,c,p},x] && ILtQ[n,0] && IntegerQ[m] && NeQ[m,-1]
```

3. $\int \frac{u}{d + d\,\text{ProductLog}[a + b\,x]}\,dx$

1: $\int \frac{1}{d + d\,\text{ProductLog}[a + b\,x]}\,dx$

Rule:

$$\int \frac{1}{d + d\,\text{ProductLog}[a + b\,x]}\,dx \longrightarrow \frac{a + b\,x}{b\,d\,\text{ProductLog}[a + b\,x]}$$

Program code:

```
Int[1/(d_+d_.*ProductLog[a_.+b_.*x_]),x_Symbol] :=
  (a+b*x)/(b*d*ProductLog[a+b*x]) /;
FreeQ[{a,b,d},x]
```

2. $\int \frac{\text{ProductLog}[a+b\,x]^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$

1. $\int \frac{\text{ProductLog}[a+b\,x]^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$ when $p>0$

1: $\int \frac{\text{ProductLog}[a+b\,x]}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$

Derivation: Algebraic simplification

Basis: $\frac{z}{1+z} == 1-\frac{1}{1+z}$

Rule:

$$\int \frac{\text{ProductLog}[a+b\,x]}{d+d\,\text{ProductLog}[a+b\,x]}\,dx \longrightarrow d\,x-\int \frac{1}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$$

Program code:

```
Int[ProductLog[a_.+b_.*x_]/(d_+d_.*ProductLog[a_.+b_.*x_]),x_Symbol] :=
  d*x - Int[1/(d+d*ProductLog[a+b*x]),x] /;
FreeQ[{a,b,d},x]
```

2: $\int \frac{(c\,\text{ProductLog}[a+b\,x])^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$ when $p>0$

Rule: If $p>0$, then

$$\int \frac{(c\,\text{ProductLog}[a+b\,x])^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx \longrightarrow \frac{c\,(a+b\,x)\,(c\,\text{ProductLog}[a+b\,x])^{p-1}}{b\,d}-c\,p\int \frac{(c\,\text{ProductLog}[a+b\,x])^{p-1}}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$$

Program code:

```
Int[(c_.*ProductLog[a_.+b_.*x_])^p_/(d_+d_.*ProductLog[a_.+b_.*x_]),x_Symbol] :=
  c*(a+b*x)*(c*ProductLog[a+b*x])^(p-1)/(b*d) -
  c*p*Int[(c*ProductLog[a+b*x])^(p-1)/(d+d*ProductLog[a+b*x]),x] /;
FreeQ[{a,b,c,d},x] && GtQ[p,0]
```

2. $\int \frac{\text{ProductLog}[a+b\,x]^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$ when $p<0$

1: $\int \frac{1}{\text{ProductLog}[a+b\,x]\,(d+d\,\text{ProductLog}[a+b\,x])}\,dx$

Rule:

$$\int \frac{1}{\text{ProductLog}[a+b\,x]\,(d+d\,\text{ProductLog}[a+b\,x])}\,dx \longrightarrow \frac{\text{ExpIntegralEi}[\text{ProductLog}[a+b\,x]]}{b\,d}$$

Program code:

```
Int[1/(ProductLog[a_.+b_.*x_]*(d_+d_.*ProductLog[a_.+b_.*x_])),x_Symbol] :=
  ExpIntegralEi[ProductLog[a+b*x]]/(b*d) /;
FreeQ[{a,b,d},x]
```

2. $\int \frac{1}{\text{Sqrt}[c\,\text{ProductLog}[a+b\,x]]\,(d+d\,\text{ProductLog}[a+b\,x])}\,dx$

1: $\int \frac{1}{\text{Sqrt}[c\,\text{ProductLog}[a+b\,x]]\,(d+d\,\text{ProductLog}[a+b\,x])}\,dx$ when $c>0$

Rule: If $c>0$, then

$$\int \frac{1}{\text{Sqrt}[c\,\text{ProductLog}[a+b\,x]]\,(d+d\,\text{ProductLog}[a+b\,x])}\,dx \to \frac{\sqrt{\pi\,c}}{b\,c\,d}\,\text{Erfi}\left[\frac{\sqrt{c\,\text{ProductLog}[a+b\,x]}}{\sqrt{c}}\right]$$

Program code:

```
Int[1/(Sqrt[c_.*ProductLog[a_.+b_.*x_]]*(d_+d_.*ProductLog[a_.+b_.*x_])),x_Symbol] :=
  Rt[Pi*c,2]*Erfi[Sqrt[c*ProductLog[a+b*x]]/Rt[c,2]]/(b*c*d) /;
FreeQ[{a,b,c,d},x] && PosQ[c]
```

2: $\int \frac{1}{\text{Sqrt}[c\,\text{ProductLog}[a+b\,x]]\,(d+d\,\text{ProductLog}[a+b\,x])}\,dx$ when $c<0$

Rule: If $c<0$, then

$$\int \frac{1}{\text{Sqrt}[c\,\text{ProductLog}[a+b\,x]]\,(d+d\,\text{ProductLog}[a+b\,x])}\,dx \to \frac{\sqrt{-\pi\,c}}{b\,c\,d}\,\text{Erf}\left[\frac{\sqrt{c\,\text{ProductLog}[a+b\,x]}}{\sqrt{-c}}\right]$$

Program code:

```
Int[1/(Sqrt[c_.*ProductLog[a_.+b_.*x_]]*(d_+d_.*ProductLog[a_.+b_.*x_])),x_Symbol] :=
  Rt[-Pi*c,2]*Erf[Sqrt[c*ProductLog[a+b*x]]/Rt[-c,2]]/(b*c*d) /;
FreeQ[{a,b,c,d},x] && NegQ[c]
```

3: $\int \frac{(c\,\text{ProductLog}[a+b\,x])^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$ when $p<-1$

Rule: If $p<-1$, then

$$\int \frac{(c\,\text{ProductLog}[a+b\,x])^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx \longrightarrow \frac{(a+b\,x)\,(c\,\text{ProductLog}[a+b\,x])^p}{b\,d\,(p+1)}-\frac{1}{c\,(p+1)}\int \frac{(c\,\text{ProductLog}[a+b\,x])^{p+1}}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$$

Program code:

```
Int[(c_.*ProductLog[a_.+b_.*x_])^p_/(d_+d_.*ProductLog[a_.+b_.*x_]),x_Symbol] :=
  (a+b*x)*(c*ProductLog[a+b*x])^p/(b*d*(p+1)) -
  1/(c*(p+1))*Int[(c*ProductLog[a+b*x])^(p+1)/(d+d*ProductLog[a+b*x]),x] /;
FreeQ[{a,b,c,d},x] && LtQ[p,-1]
```

3: $\int \frac{(c\,\text{ProductLog}[a+b\,x])^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$

Rule:

$$\int \frac{(c\,\text{ProductLog}[a+b\,x])^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx \longrightarrow \frac{\text{Gamma}[p+1,\,-\text{ProductLog}[a+b\,x]]\,(c\,\text{ProductLog}[a+b\,x])^p}{b\,d\,(-\text{ProductLog}[a+b\,x])^p}$$

Program code:

```
Int[(c_.*ProductLog[a_.+b_.*x_])^p_./(d_+d_.*ProductLog[a_.+b_.*x_]),x_Symbol] :=
  Gamma[p+1,-ProductLog[a+b*x]]*(c*ProductLog[a+b*x])^p/(b*d*(-ProductLog[a+b*x])^p) /;
FreeQ[{a,b,c,d,p},x]
```

3: $\int \frac{(e+f\,x)^m}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$ when $m\in\mathbb{Z}^+$

Derivation: Integration by substitution

Rule: If $m\in\mathbb{Z}^+$, then

$$\int \frac{(e+f\,x)^m}{d+d\,\text{ProductLog}[a+b\,x]}\,dx \to \frac{1}{b^{m+1}}\,\text{Subst}\Big[\int \frac{1}{d+d\,\text{ProductLog}[x]}\,\text{ExpandIntegrand}\big[(b\,e-a\,f+f\,x)^m,\,x\big]\,dx,\,x,\,a+b\,x\Big]$$

Program code:

```
Int[(e_.+f_.*x_)^m_./(d_+d_.*ProductLog[a_+b_.*x_]),x_Symbol] :=
  1/b^(m+1)*Subst[Int[ExpandIntegrand[1/(d+d*ProductLog[x]),(b*e-a*f+f*x)^m,x],x],x,a+b*x] /;
FreeQ[{a,b,d,e,f},x] && IGtQ[m,0]
```

4: $\int \frac{(e+f\,x)^m\,(c\,\text{ProductLog}[a+b\,x])^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx$ when $m\in\mathbb{Z}^+$

Derivation: Integration by substitution

Rule: If $m\in\mathbb{Z}^+$, then

$$\int \frac{(e+f\,x)^m\,(c\,\text{ProductLog}[a+b\,x])^p}{d+d\,\text{ProductLog}[a+b\,x]}\,dx \to \frac{1}{b^{m+1}}\,\text{Subst}\Big[\int \frac{(c\,\text{ProductLog}[x])^p}{d+d\,\text{ProductLog}[x]}\,\text{ExpandIntegrand}\big[(b\,e-a\,f+f\,x)^m,\,x\big]\,dx,\,x,\,a+b\,x\Big]$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*(c_.*ProductLog[a_+b_.*x_])^p_./(d_+d_.*ProductLog[a_+b_.*x_]),x_Symbol] :=
  1/b^(m+1)*Subst[Int[ExpandIntegrand[(c*ProductLog[x])^p/(d+d*ProductLog[x]),(b*e-a*f+f*x)^m,x],x],x,a+b*x] /;
FreeQ[{a,b,c,d,e,f,p},x] && IGtQ[m,0]
```

4. $\int \frac{u}{d + d\,\text{ProductLog}[a\,x^n]}\,dx$

1: $\int \frac{1}{d + d\,\text{ProductLog}[a\,x^n]}\,dx$ when $n \in \mathbb{Z}^-$

Derivation: Integration by substitution

Basis: $\int f[x]\,dx = -\text{Subst}\left[\int \frac{f\left[\frac{1}{x}\right]}{x^2}\,dx,\ x,\ \frac{1}{x}\right]$

Rule: If $n \in \mathbb{Z}^-$, then

$$\int \frac{1}{d + d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow -\text{Subst}\left[\int \frac{1}{x^2\,(d + d\,\text{ProductLog}[a\,x^{-n}])}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[1/(d_+d_.*ProductLog[a_.*x_^n_]),x_Symbol] :=
  -Subst[Int[1/(x^2*(d+d*ProductLog[a*x^(-n)])),x],x,1/x] /;
FreeQ[{a,d},x] && ILtQ[n,0]
```

2. $\int \frac{(c\, \text{ProductLog}[a\, x^n])^p}{d + d\, \text{ProductLog}[a\, x^n]}\, dx$

1: $\int \frac{(c\, \text{ProductLog}[a\, x^n])^p}{d + d\, \text{ProductLog}[a\, x^n]}\, dx$ when $n\,(p-1) == -1$

Rule: If $n\,(p-1) == -1$, then

$$\int \frac{(c\, \text{ProductLog}[a\, x^n])^p}{d + d\, \text{ProductLog}[a\, x^n]}\, dx \longrightarrow \frac{c\, x\, (c\, \text{ProductLog}[a\, x^n])^{p-1}}{d}$$

Program code:

```
Int[(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  c*x*(c*ProductLog[a*x^n])^(p-1)/d /;
FreeQ[{a,c,d,n,p},x] && EqQ[n*(p-1),-1]
```

3: $\int \frac{\text{ProductLog}[a\, x^n]^p}{d + d\, \text{ProductLog}[a\, x^n]}\, dx$ when $p \in \mathbb{Z} \land n\, p == -1$

Rule: If $p \in \mathbb{Z} \land n\, p == -1$, then

$$\int \frac{\text{ProductLog}[a\, x^n]^p}{d + d\, \text{ProductLog}[a\, x^n]}\, dx \longrightarrow \frac{a^p\, \text{ExpIntegralEi}[-p\, \text{ProductLog}[a\, x^n]]}{d\, n}$$

Program code:

```
Int[ProductLog[a_.*x_^n_.]^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  a^p*ExpIntegralEi[-p*ProductLog[a*x^n]]/(d*n) /;
FreeQ[{a,d},x] && IntegerQ[p] && EqQ[n*p,-1]
```

4. $\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d + d\,\text{ProductLog}[a\,x^n]}\,dx$ when $\frac{1}{n} \in \mathbb{Z} \land p == \frac{1}{2} - \frac{1}{n}$

1: $\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d + d\,\text{ProductLog}[a\,x^n]}\,dx$ when $\frac{1}{n} \in \mathbb{Z} \land p == \frac{1}{2} - \frac{1}{n} \land c\,n > 0$

Rule: If $\frac{1}{n} \in \mathbb{Z} \land p == \frac{1}{2} - \frac{1}{n} \land c\,n > 0$, then

$$\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d + d\,\text{ProductLog}[a\,x^n]}\,dx \to \frac{\sqrt{\pi\,c\,n}}{d\,n\,a^{1/n}\,c^{1/n}}\,\text{Erfi}\left[\frac{\sqrt{c\,\text{ProductLog}[a\,x^n]}}{\sqrt{c\,n}}\right]$$

Program code:

```
Int[(c_.*ProductLog[a_.*x_^n_.])^p_/(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  Rt[Pi*c*n,2]/(d*n*a^(1/n)*c^(1/n))*Erfi[Sqrt[c*ProductLog[a*x^n]]/Rt[c*n,2]] /;
FreeQ[{a,c,d},x] && IntegerQ[1/n] && EqQ[p,1/2-1/n] && PosQ[c*n]
```

2: $\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d + d\,\text{ProductLog}[a\,x^n]}\,dx$ when $\frac{1}{n} \in \mathbb{Z} \land p == \frac{1}{2} - \frac{1}{n} \land c\,n < 0$

Rule: If $\frac{1}{n} \in \mathbb{Z} \land p == \frac{1}{2} - \frac{1}{n} \land c\,n < 0$, then

$$\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d + d\,\text{ProductLog}[a\,x^n]}\,dx \to \frac{\sqrt{-\pi\,c\,n}}{d\,n\,a^{1/n}\,c^{1/n}}\,\text{Erf}\left[\frac{\sqrt{c\,\text{ProductLog}[a\,x^n]}}{\sqrt{-c\,n}}\right]$$

Program code:

```
Int[(c_.*ProductLog[a_.*x_^n_.])^p_/(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  Rt[-Pi*c*n,2]/(d*n*a^(1/n)*c^(1/n))*Erf[Sqrt[c*ProductLog[a*x^n]]/Rt[-c*n,2]] /;
FreeQ[{a,c,d},x] && IntegerQ[1/n] && EqQ[p,1/2-1/n] && NegQ[c*n]
```

5: $\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$ when $n>0\land n\,(p-1)+1>0$

Rule: If $n>0\land n\,(p-1)+1>0$, then

$$\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow \frac{c\,x\,(c\,\text{ProductLog}[a\,x^n])^{p-1}}{d}-c\,(n\,(p-1)+1)\int \frac{(c\,\text{ProductLog}[a\,x^n])^{p-1}}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$$

Program code:

```
Int[(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  c*x*(c*ProductLog[a*x^n])^(p-1)/d -
  c*(n*(p-1)+1)*Int[(c*ProductLog[a*x^n])^(p-1)/(d+d*ProductLog[a*x^n]),x] /;
FreeQ[{a,c,d},x] && GtQ[n,0] && GtQ[n*(p-1)+1,0]
```

6: $\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$ when $n>0\land n\,p+1<0$

Rule: If $n>0\land n\,p+1<0$, then

$$\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow \frac{x\,(c\,\text{ProductLog}[a\,x^n])^{p}}{d\,(n\,p+1)}-\frac{1}{c\,(n\,p+1)}\int \frac{(c\,\text{ProductLog}[a\,x^n])^{p+1}}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$$

Program code:

```
Int[(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  x*(c*ProductLog[a*x^n])^p/(d*(n*p+1)) -
  1/(c*(n*p+1))*Int[(c*ProductLog[a*x^n])^(p+1)/(d+d*ProductLog[a*x^n]),x] /;
FreeQ[{a,c,d},x] && GtQ[n,0] && LtQ[n*p+1,0]
```

7: $\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$ when $n \in \mathbb{Z}^-$

Derivation: Integration by substitution

Basis: $\int f[x]\,dx = -\text{Subst}\left[\int \frac{f\left[\frac{1}{x}\right]}{x^2}\,dx,\,x,\,\frac{1}{x}\right]$

Rule: If $n \in \mathbb{Z}^-$, then

$$\int \frac{(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx \to -\text{Subst}\left[\int \frac{(c\,\text{ProductLog}[a\,x^{-n}])^p}{x^2\,(d+d\,\text{ProductLog}[a\,x^{-n}])}\,dx,\,x,\,\frac{1}{x}\right]$$

Program code:

```
Int[(c_.*ProductLog[a_.*x_^n_])^p_./(d_+d_.*ProductLog[a_.*x_^n_]),x_Symbol] :=
  -Subst[Int[(c*ProductLog[a*x^(-n)])^p/(x^2*(d+d*ProductLog[a*x^(-n)])),x],x,1/x] /;
FreeQ[{a,c,d,p},x] && ILtQ[n,0]
```

3. $\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x]}\,dx$

1: $\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x]}\,dx$ when $m > 0$

Rule: If $m > 0$, then

$$\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x]}\,dx \longrightarrow \frac{x^{m+1}}{d\,(m+1)\,\text{ProductLog}[a\,x]} - \frac{m}{m+1}\int \frac{x^m}{\text{ProductLog}[a\,x]\,\left(d + d\,\text{ProductLog}[a\,x]\right)}\,dx$$

Program code:

```
Int[x_^m_./(d_+d_.*ProductLog[a_.*x_]),x_Symbol] :=
  x^(m+1)/(d*(m+1)*ProductLog[a*x]) -
  m/(m+1)*Int[x^m/(ProductLog[a*x]*(d+d*ProductLog[a*x])),x] /;
FreeQ[{a,d},x] && GtQ[m,0]
```

2. $\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x]}\,dx$ when $m < 0$

1: $\int \frac{1}{x\,\left(d + d\,\text{ProductLog}[a\,x]\right)}\,dx$

Rule:

$$\int \frac{1}{x\,\left(d + d\,\text{ProductLog}[a\,x]\right)}\,dx \longrightarrow \frac{\text{Log}[\text{ProductLog}[a\,x]]}{d}$$

Program code:

```
Int[1/(x_*(d_+d_.*ProductLog[a_.*x_])),x_Symbol] :=
  Log[ProductLog[a*x]]/d /;
FreeQ[{a,d},x]
```

2: $\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x]}\,dx$ when $m < -1$

Rule: If $m < -1$, then

$$\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x]}\,dx \rightarrow \frac{x^{m+1}}{d\,(m+1)} - \int \frac{x^m\,\text{ProductLog}[a\,x]}{d + d\,\text{ProductLog}[a\,x]}\,dx$$

Program code:

```
Int[x_^m_./(d_+d_.*ProductLog[a_.*x_]),x_Symbol] :=
  x^(m+1)/(d*(m+1)) -
  Int[x^m*ProductLog[a*x]/(d+d*ProductLog[a*x]),x] /;
FreeQ[{a,d},x] && LtQ[m,-1]
```

3: $\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x]}\,dx$ when $m \notin \mathbb{Z}$

Rule: If $m \notin \mathbb{Z}$, then

$$\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x]}\,dx \rightarrow \frac{x^m\,\text{Gamma}[m+1, -(m+1)\,\text{ProductLog}[a\,x]]}{a\,d\,(m+1)\,e^{m\,\text{ProductLog}[a\,x]}\,(-(m+1)\,\text{ProductLog}[a\,x])^m}$$

Program code:

```
Int[x_^m_./(d_+d_.*ProductLog[a_.*x_]),x_Symbol] :=
  x^m*Gamma[m+1,-(m+1)*ProductLog[a*x]]/
    (a*d*(m+1)*E^(m*ProductLog[a*x])*(-(m+1)*ProductLog[a*x])^m) /;
FreeQ[{a,d,m},x] && Not[IntegerQ[m]]
```

4. $\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x^n]}\,dx$

1: $\int \frac{1}{x\,(d + d\,\text{ProductLog}[a\,x^n])}\,dx$

Rule:

$$\int \frac{1}{x\,(d + d\,\text{ProductLog}[a\,x^n])}\,dx \longrightarrow \frac{\text{Log}[\text{ProductLog}[a\,x^n]]}{d\,n}$$

Program code:

```
Int[1/(x_*(d_+d_.*ProductLog[a_.*x_^n_.])),x_Symbol] :=
  Log[ProductLog[a*x^n]]/(d*n) /;
FreeQ[{a,d,n},x]
```

2: $\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x^n]}\,dx$ when $m \in \mathbb{Z} \land n \in \mathbb{Z}^- \land m \neq -1$

Derivation: Integration by substitution

Basis: $\int f[x]\,dx = -\text{Subst}\left[\int \frac{f\left[\frac{1}{x}\right]}{x^2}\,dx,\ x,\ \frac{1}{x}\right]$

Rule: If $m \in \mathbb{Z} \land n \in \mathbb{Z}^- \land m \neq -1$, then

$$\int \frac{x^m}{d + d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow -\text{Subst}\left[\int \frac{1}{x^{m+2}\,(d + d\,\text{ProductLog}[a\,x^{-n}])}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[x_^m_./(d_+d_.*ProductLog[a_.*x_^n_]),x_Symbol] :=
  -Subst[Int[1/(x^(m+2)*(d+d*ProductLog[a*x^(-n)])),x],x,1/x] /;
FreeQ[{a,d},x] && IntegerQ[m] && ILtQ[n,0] && NeQ[m,-1]
```

5. $\int \frac{x^m \left(c\,\text{ProductLog}\left[a\,x^n\right]\right)^p}{d+d\,\text{ProductLog}\left[a\,x^n\right]}\,dx$

1: $\int \frac{\left(c\,\text{ProductLog}\left[a\,x^n\right]\right)^p}{x\left(d+d\,\text{ProductLog}\left[a\,x^n\right]\right)}\,dx$

Rule:

$$\int \frac{\left(c\,\text{ProductLog}\left[a\,x^n\right]\right)^p}{x\left(d+d\,\text{ProductLog}\left[a\,x^n\right]\right)}\,dx \longrightarrow \frac{\left(c\,\text{ProductLog}\left[a\,x^n\right]\right)^p}{d\,n\,p}$$

Program code:

```
Int[(c_.*ProductLog[a_.*x_^n_.])^p_./(x_*(d_+d_.*ProductLog[a_.*x_^n_.])),x_Symbol] :=
  (c*ProductLog[a*x^n])^p/(d*n*p) /;
FreeQ[{a,c,d,n,p},x]
```

2. $\int \frac{x^m (c \, \text{ProductLog}[a \, x^n])^p}{d + d \, \text{ProductLog}[a \, x^n]} \, dx$ when $m \neq -1$

1: $\int \frac{x^m (c \, \text{ProductLog}[a \, x^n])^p}{d + d \, \text{ProductLog}[a \, x^n]} \, dx$ when $m \neq -1 \land m + n (p - 1) == -1$

Rule: If $m \neq -1 \land m + n (p - 1) == -1$, then

$$\int \frac{x^m (c \, \text{ProductLog}[a \, x^n])^p}{d + d \, \text{ProductLog}[a \, x^n]} \, dx \longrightarrow \frac{c \, x^{m+1} (c \, \text{ProductLog}[a \, x^n])^{p-1}}{d (m + 1)}$$

Program code:

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  c*x^(m+1)*(c*ProductLog[a*x^n])^(p-1)/(d*(m+1)) /;
FreeQ[{a,c,d,m,n,p},x] && NeQ[m,-1] && EqQ[m+n*(p-1),-1]
```

2: $\int \frac{x^m \, \text{ProductLog}[a \, x^n]^p}{d + d \, \text{ProductLog}[a \, x^n]} \, dx$ when $p \in \mathbb{Z} \land m + n \, p == -1$

Rule: If $p \in \mathbb{Z} \land m + n \, p == -1$, then

$$\int \frac{x^m \, \text{ProductLog}[a \, x^n]^p}{d + d \, \text{ProductLog}[a \, x^n]} \, dx \longrightarrow \frac{a^p \, \text{ExpIntegralEi}[-p \, \text{ProductLog}[a \, x^n]]}{d \, n}$$

Program code:

```
Int[x_^m_.*ProductLog[a_.*x_^n_.]^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  a^p*ExpIntegralEi[-p*ProductLog[a*x^n]]/(d*n) /;
FreeQ[{a,d,m,n},x] && IntegerQ[p] && EqQ[m+n*p,-1]
```

3. $\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$ when $m\neq -1\land p-\frac{1}{2}\in\mathbb{Z}\land m+n\left(p-\frac{1}{2}\right)+1==0$

1: $\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$ when $m\neq -1\land p-\frac{1}{2}\in\mathbb{Z}\land m+n\left(p-\frac{1}{2}\right)==-1\land \frac{c}{p-\frac{1}{2}}>0$

Rule: If $m\neq -1\land p-\frac{1}{2}\in\mathbb{Z}\land m+n\left(p-\frac{1}{2}\right)==-1\land \frac{c}{p-\frac{1}{2}}>0$, then

$$\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx\to\frac{a^{p-\frac{1}{2}}\,c^{p-\frac{1}{2}}}{d\,n}\sqrt{\frac{\pi\,c}{p-\frac{1}{2}}}\,\text{Erf}\left[\frac{\sqrt{c\,\text{ProductLog}[a\,x^n]}}{\sqrt{\frac{c}{p-\frac{1}{2}}}}\right]$$

Program code:

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_/(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  a^(p-1/2)*c^(p-1/2)*Rt[Pi*c/(p-1/2),2]*Erf[Sqrt[c*ProductLog[a*x^n]]/Rt[c/(p-1/2),2]]/(d*n) /;
FreeQ[{a,c,d,m,n},x] && NeQ[m,-1] && IntegerQ[p-1/2] && EqQ[m+n*(p-1/2),-1] && PosQ[c/(p-1/2)]
```

2: $\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$ when $m\neq -1\land p-\frac{1}{2}\in\mathbb{Z}\land m+n\left(p-\frac{1}{2}\right)==-1\land \frac{c}{p-\frac{1}{2}}<0$

Rule: If $m\neq -1\land p-\frac{1}{2}\in\mathbb{Z}\land m+n\left(p-\frac{1}{2}\right)==-1\land \frac{c}{p-\frac{1}{2}}<0$, then

$$\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx\to\frac{a^{p-\frac{1}{2}}\,c^{p-\frac{1}{2}}}{d\,n}\sqrt{-\frac{\pi\,c}{p-\frac{1}{2}}}\,\text{Erfi}\left[\frac{\sqrt{c\,\text{ProductLog}[a\,x^n]}}{\sqrt{-\frac{c}{p-\frac{1}{2}}}}\right]$$

Program code:

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_/(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  a^(p-1/2)*c^(p-1/2)*Rt[-Pi*c/(p-1/2),2]*Erfi[Sqrt[c*ProductLog[a*x^n]]/Rt[-c/(p-1/2),2]]/(d*n) /;
FreeQ[{a,c,d,m,n},x] && NeQ[m,-1] && IntegerQ[p-1/2] && EqQ[m+n*(p-1/2),-1] && NegQ[c/(p-1/2)]
```

4: $\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$ when $m \neq -1 \land p+\frac{m+1}{n} > 1$

Rule: If $m \neq -1 \land p+\frac{m+1}{n} > 1$, then

$$\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow \frac{c\,x^{m+1}\,(c\,\text{ProductLog}[a\,x^n])^{p-1}}{d\,(m+1)} - \frac{c\,(m+n\,(p-1)+1)}{m+1}\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^{p-1}}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$$

Program code:

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  c*x^(m+1)*(c*ProductLog[a*x^n])^(p-1)/(d*(m+1)) -
  c*(m+n*(p-1)+1)/(m+1)*Int[x^m*(c*ProductLog[a*x^n])^(p-1)/(d+d*ProductLog[a*x^n]),x] /;
FreeQ[{a,c,d,m,n,p},x] && NeQ[m,-1] && GtQ[Simplify[p+(m+1)/n],1]
```

5: $\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$ when $m \neq -1 \land p+\frac{m+1}{n} < 0$

Rule: If $m \neq -1 \land p+\frac{m+1}{n} < 0$, then

$$\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^p}{d+d\,\text{ProductLog}[a\,x^n]}\,dx \longrightarrow \frac{x^{m+1}\,(c\,\text{ProductLog}[a\,x^n])^{p}}{d\,(m+n\,p+1)} - \frac{m+1}{c\,(m+n\,p+1)}\int \frac{x^m\,(c\,\text{ProductLog}[a\,x^n])^{p+1}}{d+d\,\text{ProductLog}[a\,x^n]}\,dx$$

Program code:

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  x^(m+1)*(c*ProductLog[a*x^n])^p/(d*(m+n*p+1)) -
  (m+1)/(c*(m+n*p+1))*Int[x^m*(c*ProductLog[a*x^n])^(p+1)/(d+d*ProductLog[a*x^n]),x] /;
FreeQ[{a,c,d,m,n,p},x] && NeQ[m,-1] && LtQ[Simplify[p+(m+1)/n],0]
```

6: $\int \frac{x^m (c\,\text{ProductLog}[a\,x])^p}{d + d\,\text{ProductLog}[a\,x]}\,dx$ when $m \neq -1$

Rule: If $m \neq -1$, then

$$\int \frac{x^m (c\,\text{ProductLog}[a\,x])^p}{d + d\,\text{ProductLog}[a\,x]}\,dx \to \frac{x^m\,\text{Gamma}[m+p+1, -(m+1)\,\text{ProductLog}[a\,x]]\,(c\,\text{ProductLog}[a\,x])^p}{a\,d\,(m+1)\,e^{m\,\text{ProductLog}[a\,x]}\,(-(m+1)\,\text{ProductLog}[a\,x])^{m+p}}$$

Program code:

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_])^p_./(d_+d_.*ProductLog[a_.*x_]),x_Symbol] :=
  x^m*Gamma[m+p+1,-(m+1)*ProductLog[a*x]]*(c*ProductLog[a*x])^p/
    (a*d*(m+1)*E^(m*ProductLog[a*x])*(-(m+1)*ProductLog[a*x])^(m+p)) /;
FreeQ[{a,c,d,m,p},x] && NeQ[m,-1]
```

7: $\int \frac{x^m (c\,\text{ProductLog}[a\,x^n])^p}{d + d\,\text{ProductLog}[a\,x^n]}\,dx$ when $m \neq -1 \land m \in \mathbb{Z} \land n \in \mathbb{Z}^-$

Derivation: Integration by substitution

Basis: $\int f[x]\,dx = -\text{Subst}\left[\int \frac{f\left[\frac{1}{x}\right]}{x^2}\,dx,\ x,\ \frac{1}{x}\right]$

Rule: If $m \neq -1 \land m \in \mathbb{Z} \land n \in \mathbb{Z}^-$, then

$$\int \frac{x^m (c\,\text{ProductLog}[a\,x^n])^p}{d + d\,\text{ProductLog}[a\,x^n]}\,dx \to -\text{Subst}\left[\int \frac{(c\,\text{ProductLog}[a\,x^{-n}])^p}{x^{m+2}\,(d + d\,\text{ProductLog}[a\,x^{-n}])}\,dx,\ x,\ \frac{1}{x}\right]$$

Program code:

```
Int[x_^m_.*(c_.*ProductLog[a_.*x_^n_.])^p_./(d_+d_.*ProductLog[a_.*x_^n_.]),x_Symbol] :=
  -Subst[Int[(c*ProductLog[a*x^(-n)])^p/(x^(m+2)*(d+d*ProductLog[a*x^(-n)])),x],x,1/x] /;
FreeQ[{a,c,d,p},x] && NeQ[m,-1] && IntegerQ[m] && LtQ[n,0]
```

5: $\int f[\text{ProductLog}[x]]\,dx$

Author: Rob Corless 2009-07-10

Derivation: Legendre substitution for inverse functions

Basis: $f[\text{ProductLog}[x]] == (\text{ProductLog}[z]+1)\,e^{\text{ProductLog}[z]}\,f[\text{ProductLog}[x]]\,\text{ProductLog}'[z]$

Rule:

$$\int f[\text{ProductLog}[x]]\,dx \longrightarrow \text{Subst}\left[\int (x+1)\,e^x\,f[x]\,dx,\ x,\ \text{ProductLog}[x]\right]$$

Program code:

```
Int[u_,x_Symbol] :=
  Subst[Int[SimplifyIntegrand[(x+1)*E^x*SubstFor[ProductLog[x],u,x],x],x],x,ProductLog[x]] /;
FunctionOfQ[ProductLog[x],u,x]
```