

## Rules for integrands of the form $(a + b \sin[c + d x])^n$

$$1. \int (b \sin[c + d x])^n dx$$

$$1. \int (b \sin[c + d x])^n dx \text{ when } 2n \in \mathbb{Z}$$

$$1. \int (b \sin[c + d x])^n dx \text{ when } n > 1$$

$$1: \int \sin[c + d x]^n dx \text{ when } \frac{n-1}{2} \in \mathbb{Z}^+$$

Derivation: Integration by substitution

Basis: If  $\frac{n-1}{2} \in \mathbb{Z}$ , then  $\sin[c + d x]^n = -\frac{1}{d} \text{Subst}\left[\left(1 - x^2\right)^{\frac{n-1}{2}}, x, \cos[c + d x]\right] \partial_x \cos[c + d x]$

Rule: If  $\frac{n-1}{2} \in \mathbb{Z}^+$ , then

$$\int \sin[c + d x]^n dx \rightarrow -\frac{1}{d} \text{Subst}\left[\int \left(1 - x^2\right)^{\frac{n-1}{2}} dx, x, \cos[c + d x]\right]$$

Program code:

```
Int[sin[c_+d_.*x_]^n_,x_Symbol] :=
  -1/d*Subst[Int[Expand[(1-x^2)^(n-1)/2],x],x,Cos[c+d*x]] /;
FreeQ[{c,d},x] && IGtQ[(n-1)/2,0]
```

$$2. \int (b \sin[c + dx])^n dx \text{ when } n > 1$$

$$1: \int \sin[c + dx]^2 dx$$

Derivation: Algebraic expansion

$$\text{Basis: } \sin[z]^2 = \frac{1}{2} - \frac{\cos[2z]}{2}$$

Rule:

$$\int \sin[c + dx]^2 dx \rightarrow \frac{x}{2} - \frac{\sin[2c + 2dx]}{4d}$$

Program code:

```
Int[sin[c_+d_.x_/2]^2,x_Symbol] :=
  x/2 - Sin[2*c+d*x]/(2*d) /;
FreeQ[{c,d},x]
```

$$2: \int (b \sin[c + dx])^n dx \text{ when } n > 1$$

Reference: G&R 2.510.2 with  $q \rightarrow 0$ , CRC 299

Reference: G&R 2.510.5 with  $p \rightarrow 0$ , CRC 305

Derivation: Sine recurrence 3a with  $A \rightarrow 0$ ,  $B \rightarrow a$ ,  $C \rightarrow b$ ,  $m \rightarrow m - 1$ ,  $n \rightarrow -1$

Derivation: Sine recurrence 1b with  $A \rightarrow 0$ ,  $B \rightarrow 0$ ,  $C \rightarrow b$ ,  $a \rightarrow 0$ ,  $m \rightarrow -1$ ,  $n \rightarrow n - 1$

Rule: If  $n > 1$ , then

$$\int (b \sin[c+dx])^n dx \rightarrow -\frac{b \cos[c+dx] (b \sin[c+dx])^{n-1}}{dn} + \frac{b^2 (n-1)}{n} \int (b \sin[c+dx])^{n-2} dx$$

Program code:

```
Int[(b_.sin[c_.+d_.x_])^n_,x_Symbol] :=
(* -Cot[c+d*x]*(c*sin[c+d*x])^n/(d*n) + b^2*(n-1)/n*Int[(b*sin[c+d*x])^(n-2),x] *)
-b*cos[c+d*x]*(b*sin[c+d*x])^(n-1)/(d*n) + b^2*(n-1)/n*Int[(b*sin[c+d*x])^(n-2),x] /;
FreeQ[{b,c,d},x] && GtQ[n,1] && IntegerQ[2*n]
```

**2:**  $\int (b \sin[c+dx])^n dx$  when  $n < -1$

Reference: G&R 2.510.3 with  $q \rightarrow 0$ , CRC 309

Reference: G&R 2.510.6 with  $p \rightarrow 0$ , CRC 313

Reference: G&R 2.552.3

Derivation: Sine recurrence 3a with  $A \rightarrow 0$ ,  $B \rightarrow a$ ,  $C \rightarrow b$ ,  $m \rightarrow m-1$ ,  $n \rightarrow -1$  inverted

Derivation: Sine recurrence 2a with  $A \rightarrow 1$ ,  $B \rightarrow 0$ ,  $C \rightarrow 0$ ,  $a \rightarrow 0$ ,  $m \rightarrow 0$

Rule: If  $n < -1$ , then

$$\int (b \sin[c+dx])^n dx \rightarrow \frac{\cos[c+dx] (b \sin[c+dx])^{n+1}}{bd(n+1)} + \frac{n+2}{b^2(n+1)} \int (b \sin[c+dx])^{n+2} dx$$

Program code:

```
Int[(b_.sin[c_.+d_.x_])^n_,x_Symbol] :=
Cos[c+d*x]*(b*sin[c+d*x])^(n+1)/(b*d*(n+1)) +
(n+2)/(b^2*(n+1))*Int[(b*sin[c+d*x])^(n+2),x] /;
FreeQ[{b,c,d},x] && LtQ[n,-1] && IntegerQ[2*n]
```

$$\mathbf{3:} \int \mathbf{Sin[c + d x]} \, dx$$

Reference: G&R 2.01.5, CRC 290, A&S 4.3.113

Reference: G&R 2.01.6, CRC 291, A&S 4.3.114

Derivation: Primitive rule

Basis:  $\partial_x \text{Cos}[c + d x] = -d \text{Sin}[c + d x]$

Rule:

$$\int \mathbf{Sin[c + d x]} \, dx \rightarrow -\frac{\mathbf{Cos[c + d x]}}{d}$$

Program code:

```
Int[sin[c_.+Pi/2+d_.*x_],x_Symbol] :=
  Sin[c+d*x]/d /;
FreeQ[{c,d},x]
```

```
Int[sin[c_.+d_.*x_],x_Symbol] :=
  -Cos[c+d*x]/d /;
FreeQ[{c,d},x]
```

$$\mathbf{x:} \int \frac{1}{\mathbf{Sin[c + d x]}} \, dx$$

Note: This rule not necessary since *Mathematica* automatically simplifies  $\frac{1}{\text{Sin}[z]}$  to  $\text{Csc}[z]$ .

Rule:

$$\int \frac{1}{\sin[c+dx]} dx \rightarrow \int \operatorname{Csc}[c+dx] dx$$

Program code:

```
(* Int[1/sin[c_.+d_.*x_],x_Symbol] :=
  Int[Csc[c+d*x],x] /;
FreeQ[{c,d},x] *)
```

$$4. \int \sqrt{b \sin[c+dx]} dx$$

$$1: \int \sqrt{\sin[c+dx]} dx$$

Derivation: Primitive rule

$$\text{Basis: } \partial_x \operatorname{EllipticE}\left[\frac{1}{2} \left(x - \frac{\pi}{2}\right), 2\right] = \frac{\sqrt{\sin[x]}}{2}$$

Rule:

$$\int \sqrt{\sin[c+dx]} dx \rightarrow \frac{2}{d} \operatorname{EllipticE}\left[\frac{1}{2} \left(c - \frac{\pi}{2} + dx\right), 2\right]$$

Program code:

```
Int[Sqrt[sin[c_.+d_.*x_]],x_Symbol] :=
  2/d*EllipticE[1/2*(c-Pi/2+d*x),2] /;
FreeQ[{c,d},x]
```

$$2: \int \sqrt{b \sin[c+dx]} dx$$

Derivation: Piecewise constant extraction

Basis:  $\partial_x \frac{\sqrt{b \sin[c+dx]}}{\sqrt{\sin[c+dx]}} == 0$

Rule:

$$\int \sqrt{b \sin[c+dx]} \, dx \rightarrow \frac{\sqrt{b \sin[c+dx]}}{\sqrt{\sin[c+dx]}} \int \sqrt{\sin[c+dx]} \, dx$$

Program code:

```
Int[Sqrt[b_*sin[c_.+d.*x_]],x_Symbol] :=
  Sqrt[b*Sin[c+d*x]]/Sqrt[Sin[c+d*x]]*Int[Sqrt[Sin[c+d*x]],x] /;
FreeQ[{b,c,d},x]
```

$$5. \int \frac{1}{\sqrt{b \sin[c+dx]}} dx$$

$$1: \int \frac{1}{\sqrt{\sin[c+dx]}} dx$$

Derivation: Primitive rule

$$\text{Basis: } \partial_x \text{EllipticF}\left[\frac{1}{2} \left(x - \frac{\pi}{2}\right), 2\right] = \frac{1}{2\sqrt{\sin[x]}}$$

Rule:

$$\int \frac{1}{\sqrt{\sin[c+dx]}} dx \rightarrow \frac{2}{d} \text{EllipticF}\left[\frac{1}{2} \left(c - \frac{\pi}{2} + dx\right), 2\right]$$

Program code:

```
Int[1/Sqrt[sin[c_+d_.x_]],x_Symbol] :=
  2/d*EllipticF[1/2*(c-Pi/2+d*x),2] /;
FreeQ[{c,d},x]
```

$$2: \int \frac{1}{\sqrt{b \sin[c+dx]}} dx$$

Derivation: Piecewise constant extraction

$$\text{Basis: } \partial_x \frac{\sqrt{\sin[c+dx]}}{\sqrt{b \sin[c+dx]}} = 0$$

Rule:

$$\int \frac{1}{\sqrt{b \sin[c+dx]}} dx \rightarrow \frac{\sqrt{\sin[c+dx]}}{\sqrt{b \sin[c+dx]}} \int \frac{1}{\sqrt{\sin[c+dx]}} dx$$

Program code:

```
Int[1/Sqrt[b_*sin[c_+d_.*x_]],x_Symbol] :=
  Sqrt[Sin[c+d*x]]/Sqrt[b*Sin[c+d*x]]*Int[1/Sqrt[Sin[c+d*x]],x] /;
FreeQ[{b,c,d},x]
```

**2:**  $\int (b \sin[c+dx])^n dx$  when  $2n \notin \mathbb{Z}$

Derivation: Piecewise constant extraction and integration by substitution

Basis:  $\partial_x \frac{\cos[c+dx]}{\sqrt{\cos[c+dx]^2}} == 0$

Basis:  $\frac{\cos[c+dx]}{\sqrt{\cos[c+dx]^2}} \frac{\cos[c+dx]}{\sqrt{1-\sin[c+dx]^2}} == 1$

Basis:  $\cos[c+dx] F[b \sin[c+dx]] == \frac{1}{bd} \text{Subst}[F[x], x, b \sin[c+dx]] \partial_x (b \sin[c+dx])$

Note: If  $3n \in \mathbb{Z} \wedge -1 < n < 1$ , integration of  $\frac{x^n}{\sqrt{1-\frac{x^2}{b^2}}}$  results in a complicated antiderivative involving elliptic integrals

and the imaginary unit.

Rule: If  $2n \notin \mathbb{Z}$ , then

$$\int (b \sin[c+dx])^n dx \rightarrow \frac{\cos[c+dx]}{\sqrt{\cos[c+dx]^2}} \int \frac{\cos[c+dx] (b \sin[c+dx])^n}{\sqrt{1-\sin[c+dx]^2}} dx$$



$$\rightarrow \frac{\cos[c+dx]}{bd \sqrt{\cos[c+dx]^2}} \text{Subst} \left[ \int \frac{x^n}{\sqrt{1-\frac{x^2}{b^2}}} dx, x, b \sin[c+dx] \right]$$

$$\rightarrow \frac{\cos[c+dx] (b \sin[c+dx])^{n+1}}{bd(n+1) \sqrt{\cos[c+dx]^2}} \text{Hypergeometric2F1} \left[ \frac{1}{2}, \frac{n+1}{2}, \frac{n+3}{2}, \sin[c+dx]^2 \right]$$

Alternate rule: If  $2n \notin \mathbb{Z}$ , then

$$\int (b \sin[c+dx])^n dx \rightarrow - \frac{\cos[c+dx] (b \sin[c+dx])^{n+1}}{bd (\sin[c+dx]^2)^{\frac{n+1}{2}}} \text{Hypergeometric2F1} \left[ \frac{1}{2}, \frac{1-n}{2}, \frac{3}{2}, \cos[c+dx]^2 \right]$$

Program code:

```
(* Int[(b_.sin[c_.+d_.x_])^n_,x_Symbol] :=
  Cos[c+d*x]/(b*d*Sqrt[Cos[c+d*x]^2])*Subst[Int[x^n/Sqrt[1-x^2/b^2],x],x,b*Sin[c+d*x]] /;
FreeQ[{b,c,d,n},x] && Not[IntegerQ[2*n]] *)
```

```
Int[(b_.sin[c_.+d_.x_])^n_,x_Symbol] :=
  Cos[c+d*x]*(b*Sin[c+d*x])^(n+1)/(b*d*(n+1)*Sqrt[Cos[c+d*x]^2])*Hypergeometric2F1[1/2,(n+1)/2,(n+3)/2,Sin[c+d*x]^2] /;
FreeQ[{b,c,d,n},x] && Not[IntegerQ[2*n]]
```

**2:**  $\int (a + b \sin[c + dx])^2 dx$

Derivation: Algebraic expansion

Basis:  $(a + bz)^2 = \frac{1}{2} (2a^2 + b^2) + 2abz - \frac{1}{2} b^2 (1 - 2z^2)$

Rule:

$$\int (a + b \sin[c + dx])^2 dx \rightarrow \frac{(2a^2 + b^2)x}{2} - \frac{2ab \cos[c + dx]}{d} - \frac{b^2 \cos[c + dx] \sin[c + dx]}{2d}$$

Program code:

```
Int[(a+b_.sin[c_.+d_.x_])^2,x_Symbol] :=
  (2*a^2+b^2)*x/2 - 2*a*b*cos[c+d*x]/d - b^2*cos[c+d*x]*sin[c+d*x]/(2*d) /;
FreeQ[{a,b,c,d},x]
```

3.  $\int (a + b \sin[c + dx])^n dx$  when  $a^2 - b^2 = 0$

1.  $\int (a + b \sin[c + dx])^n dx$  when  $a^2 - b^2 = 0 \wedge 2n \in \mathbb{Z}$

1.  $\int (a + b \sin[c + dx])^n dx$  when  $a^2 - b^2 = 0 \wedge 2n \in \mathbb{Z}^+$

**1:**  $\int (a + b \sin[c + dx])^n dx$  when  $a^2 - b^2 = 0 \wedge n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If  $a^2 - b^2 = 0 \wedge n \in \mathbb{Z}^+$ , then

$$\int (a + b \sin[c + dx])^n dx \rightarrow \int \text{ExpandTrig}[(a + b \sin[c + dx])^n, x] dx$$

Program code:

```
Int[(a+b_.sin[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[ExpandTrig[(a+b*sin[c+d*x])^n,x],x] /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2-b^2,0] && IGtQ[n,0]
```

$$2. \int (a + b \sin[c + dx])^n dx \text{ when } a^2 - b^2 = 0 \wedge n + \frac{1}{2} \in \mathbb{Z}^+$$

$$1: \int \sqrt{a + b \sin[c + dx]} dx \text{ when } a^2 - b^2 = 0$$

Derivation: Singly degenerate sine recurrence 1b with  $A \rightarrow c$ ,  $B \rightarrow d$ ,  $m \rightarrow \frac{1}{2}$ ,  $n \rightarrow -1$ ,  $p \rightarrow 0$

Rule: If  $a^2 - b^2 = 0$ , then

$$\int \sqrt{a + b \sin[c + dx]} dx \rightarrow -\frac{2 b \cos[c + dx]}{d \sqrt{a + b \sin[c + dx]}}$$

Program code:

```
Int[Sqrt[a_+b_.sin[c_+d_.x_]],x_Symbol] :=
  -2*b*cos[c+d*x]/(d*Sqrt[a+b*sin[c+d*x]]) /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0]
```

**2:**  $\int (a + b \sin[c + dx])^n dx$  when  $a^2 - b^2 = 0 \wedge n - \frac{1}{2} \in \mathbb{Z}^+$

Reference: G&R 2.555.? inverted

Derivation: Singly degenerate sine recurrence 1b with  $A \rightarrow c$ ,  $B \rightarrow d$ ,  $n \rightarrow -1$ ,  $p \rightarrow 0$

Rule: If  $a^2 - b^2 = 0 \wedge n - \frac{1}{2} \in \mathbb{Z}^+$ , then

$$\int (a + b \sin[c + dx])^n dx \rightarrow -\frac{b \cos[c + dx] (a + b \sin[c + dx])^{n-1}}{dn} + \frac{a(2n-1)}{n} \int (a + b \sin[c + dx])^{n-1} dx$$

Program code:

```
Int[(a_+b_.sin[c_+d_.x_])^n_,x_Symbol] :=
  -b*cos[c+d*x]*(a+b*sin[c+d*x])^(n-1)/(d*n) +
  a*(2*n-1)/n*Int[(a+b*sin[c+d*x])^(n-1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0] && IGtQ[n-1/2,0]
```

$$2. \int (a + b \sin[c + dx])^n dx \text{ when } a^2 - b^2 = 0 \wedge 2n \in \mathbb{Z}^-$$

$$1: \int \frac{1}{a + b \sin[c + dx]} dx \text{ when } a^2 - b^2 = 0$$

Reference: G&R 2.555.3', CRC 337', A&S 4.3.134'/5'

Derivation: Singly degenerate sine recurrence 2a with  $A \rightarrow 1$ ,  $B \rightarrow 0$ ,  $m \rightarrow -1$ ,  $n \rightarrow 0$ ,  $p \rightarrow 0$

Rule: If  $a^2 - b^2 = 0$ , then

$$\int \frac{1}{a + b \sin[c + dx]} dx \rightarrow -\frac{\cos[c + dx]}{d (b + a \sin[c + dx])}$$

Program code:

```
Int[1/(a+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  -Cos[c+d*x]/(d*(b+a*Sin[c+d*x])) /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0]
```

**2:**  $\int \frac{1}{\sqrt{a+b \sin[c+dx]}} dx$  when  $a^2 - b^2 = 0$

Derivation: Integration by substitution

Basis: If  $a^2 - b^2 = 0$ , then  $\frac{1}{\sqrt{a+b \sin[c+dx]}} = -\frac{2}{d} \text{Subst} \left[ \frac{1}{2a-x^2}, x, \frac{b \cos[c+dx]}{\sqrt{a+b \sin[c+dx]}} \right] \partial_x \frac{b \cos[c+dx]}{\sqrt{a+b \sin[c+dx]}}$

Rule: If  $a^2 - b^2 = 0$ , then

$$\int \frac{1}{\sqrt{a+b \sin[c+dx]}} dx \rightarrow -\frac{2}{d} \text{Subst} \left[ \int \frac{1}{2a-x^2} dx, x, \frac{b \cos[c+dx]}{\sqrt{a+b \sin[c+dx]}} \right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*Sin[c_+d_.*x_]],x_Symbol] :=
  -2/d*Subst[Int[1/(2*a-x^2),x],x,b*Cos[c+d*x]/Sqrt[a+b*Sin[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0]
```

$$\mathbf{3:} \int (a + b \sin[c + dx])^n dx \text{ when } a^2 - b^2 = 0 \wedge n < -1 \wedge 2n \in \mathbb{Z}$$

Reference: G&R 2.555.?

Derivation: Singly degenerate sine recurrence 2a with  $A \rightarrow 1$ ,  $B \rightarrow 0$ ,  $n \rightarrow 0$ ,  $p \rightarrow 0$

Rule: If  $a^2 - b^2 = 0 \wedge n < -1 \wedge 2n \in \mathbb{Z}$ , then

$$\int (a + b \sin[c + dx])^n dx \rightarrow \frac{b \cos[c + dx] (a + b \sin[c + dx])^n}{a d (2n + 1)} + \frac{n + 1}{a (2n + 1)} \int (a + b \sin[c + dx])^{n+1} dx$$

Program code:

```
Int[(a_+b_.sin[c_+d_.x_])^n_,x_Symbol] :=
  b*cos[c+d*x]*(a+b*sin[c+d*x])^n/(a*d*(2*n+1)) +
  (n+1)/(a*(2*n+1))*Int[(a+b*sin[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0] && LtQ[n,-1] && IntegerQ[2*n]
```

$$2. \int (a + b \sin[c + dx])^n dx \text{ when } a^2 - b^2 = 0 \wedge 2n \notin \mathbb{Z}$$

$$\mathbf{x:} \int (a + b \sin[c + dx])^n dx \text{ when } a^2 - b^2 = 0 \wedge 2n \notin \mathbb{Z}$$

Derivation: Piecewise constant extraction and integration by substitution

$$\text{Basis: If } a^2 - b^2 = 0, \text{ then } \partial_x \frac{\cos[c+dx]}{\sqrt{a-b \sin[c+dx]} \sqrt{a+b \sin[c+dx]}} = 0$$

$$\text{Basis: If } a^2 - b^2 = 0, \text{ then } \frac{a^2 \cos[c+dx]}{\sqrt{a+b \sin[c+dx]} \sqrt{a-b \sin[c+dx]}} - \frac{\cos[c+dx]}{\sqrt{a+b \sin[c+dx]} \sqrt{a-b \sin[c+dx]}} = 1$$

$$\text{Basis: } \cos[c + dx] F[\sin[c + dx]] = \frac{1}{d} \text{Subst}[F[x], x, \sin[c + dx]] \partial_x \sin[c + dx]$$

Note: If  $3n \in \mathbb{Z}$ , this results in a complicated expression involving elliptic integrals instead of a single hypergeometric



function.

Rule: If  $a^2 - b^2 \neq 0 \wedge 2n \notin \mathbb{Z}$ , then

$$\int (a + b \sin[c + dx])^n dx \rightarrow$$

$$\frac{a^2 \cos[c + dx]}{\sqrt{a + b \sin[c + dx]} \sqrt{a - b \sin[c + dx]}} \int \frac{\cos[c + dx] (a + b \sin[c + dx])^{n-\frac{1}{2}}}{\sqrt{a - b \sin[c + dx]}} dx \rightarrow$$

$$\frac{a^2 \cos[c + dx]}{d \sqrt{a + b \sin[c + dx]} \sqrt{a - b \sin[c + dx]}} \text{Subst} \left[ \int \frac{(a + bx)^{n-\frac{1}{2}}}{\sqrt{a - bx}} dx, x, \sin[c + dx] \right]$$

Program code:

```
(* Int[(a_+b_.sin[c_+d_.x_])^n_,x_Symbol] :=
  a^2*cos[c+d*x]/(d*Sqrt[a+b*sin[c+d*x]]*Sqrt[a-b*sin[c+d*x]])*Subst[Int[(a+b*x)^(n-1/2)/Sqrt[a-b*x],x],x,Sin[c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[2*n]] *)
```

**1:**  $\int (a + b \sin[c + dx])^n dx$  when  $a^2 - b^2 = 0 \wedge 2n \notin \mathbb{Z} \wedge a > 0$

Derivation: Piecewise constant extraction and integration by substitution

Rule: If  $a^2 - b^2 = 0 \wedge 2n \notin \mathbb{Z} \wedge a > 0$ , then

$$\int (a + b \sin[c + dx])^n dx \rightarrow -\frac{2^{n+\frac{1}{2}} a^{n-\frac{1}{2}} b \cos[c + dx]}{d \sqrt{a + b \sin[c + dx]}} \text{Hypergeometric2F1}\left[\frac{1}{2}, \frac{1}{2} - n, \frac{3}{2}, \frac{1}{2} \left(1 - \frac{b \sin[c + dx]}{a}\right)\right]$$

Program code:

```
Int[(a_+b_.sin[c_+d_.x_])^n_,x_Symbol] :=
  -2^(n+1/2)*a^(n-1/2)*b*cos[c+d*x]/(d*Sqrt[a+b*sin[c+d*x]])*Hypergeometric2F1[1/2,1/2-n,3/2,1/2*(1-b*sin[c+d*x]/a)] /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[2*n]] && GtQ[a,0]
```

**2:**  $\int (a + b \sin[c + dx])^n dx$  when  $a^2 - b^2 = 0 \wedge 2n \notin \mathbb{Z} \wedge a \neq 0$

Derivation: Piecewise constant extraction

Basis:  $\partial_x \frac{(a+b \sin[c+dx])^n}{\left(1+\frac{b}{a} \sin[c+dx]\right)^n} = 0$

Rule: If  $a^2 - b^2 = 0 \wedge 2n \notin \mathbb{Z} \wedge a \neq 0$ , then

$$\int (a + b \sin[c + dx])^n dx \rightarrow \frac{a^{\text{IntPart}[n]} (a + b \sin[c + dx])^{\text{FracPart}[n]}}{\left(1 + \frac{b}{a} \sin[c + dx]\right)^{\text{FracPart}[n]}} \int \left(1 + \frac{b}{a} \sin[c + dx]\right)^n dx$$

Program code:

```
Int[(a_+b_.sin[c_+d_.x_])^n_,x_Symbol] :=
  a^IntPart[n]*(a+b*sin[c+d*x])^FracPart[n]/(1+b/a*sin[c+d*x])^FracPart[n]*Int[(1+b/a*sin[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[2*n]] && Not[GtQ[a,0]]
```

4.  $\int (a + b \sin[c + dx])^n dx$  when  $a^2 - b^2 \neq 0$

1.  $\int (a + b \sin[c + dx])^n dx$  when  $a^2 - b^2 \neq 0 \wedge 2n \in \mathbb{Z}$

1.  $\int (a + b \sin[c + dx])^n dx$  when  $a^2 - b^2 \neq 0 \wedge 2n \in \mathbb{Z}^+$

1.  $\int \sqrt{a + b \sin[c + dx]} dx$  when  $a^2 - b^2 \neq 0$

**1:**  $\int \sqrt{a + b \sin[c + dx]} dx$  when  $a^2 - b^2 \neq 0 \wedge a + b > 0$

Derivation: Primitive rule

Basis: If  $a + b > 0$ , then  $\partial_x \text{EllipticE}\left[\frac{1}{2}\left(x - \frac{\pi}{2}\right), \frac{2b}{a+b}\right] = \frac{1}{2\sqrt{a+b}} \sqrt{a + b \sin[x]}$

Rule: If  $a^2 - b^2 \neq 0 \wedge a + b > 0$ , then

$$\int \sqrt{a + b \sin[c + dx]} dx \rightarrow \frac{2\sqrt{a+b}}{d} \text{EllipticE}\left[\frac{1}{2}\left(c - \frac{\pi}{2} + dx\right), \frac{2b}{a+b}\right]$$

Program code:

```
Int[Sqrt[a_+b_.sin[c_+d.*x_]],x_Symbol] :=
  2*Sqrt[a+b]/d*EllipticE[1/2*(c-Pi/2+d*x),2*b/(a+b)] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && GtQ[a+b,0]
```

$$\text{2: } \int \sqrt{a+b \sin[c+dx]} \, dx \text{ when } a^2 - b^2 \neq 0 \wedge a-b > 0$$

Derivation: Primitive rule

$$\text{Basis: If } a-b > 0, \text{ then } \partial_x \text{EllipticE}\left[\frac{1}{2}\left(x + \frac{\pi}{2}\right), -\frac{2b}{a-b}\right] == \frac{1}{2\sqrt{a-b}} \sqrt{a+b \sin[x]}$$

Rule: If  $a^2 - b^2 \neq 0 \wedge a-b > 0$ , then

$$\int \sqrt{a+b \sin[c+dx]} \, dx \rightarrow \frac{2\sqrt{a-b}}{d} \text{EllipticE}\left[\frac{1}{2}\left(c + \frac{\pi}{2} + dx\right), -\frac{2b}{a-b}\right]$$

Program code:

```
Int[Sqrt[a_+b_.sin[c_+d_.x_]],x_Symbol] :=
  2*Sqrt[a-b]/d*EllipticE[1/2*(c+Pi/2+d*x),-2*b/(a-b)] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && GtQ[a-b,0]
```

$$\text{3: } \int \sqrt{a+b \sin[c+dx]} \, dx \text{ when } a^2 - b^2 \neq 0 \wedge a+b \neq 0$$

Derivation: Piecewise constant extraction

$$\text{Basis: } \partial_x \frac{\sqrt{a+b f[x]}}{\sqrt{\frac{a+b f[x]}{a+b}}} == 0$$

Note: Since  $\frac{a}{a+b} + \frac{b}{a+b} == 1 > 0$ , the above rule applies to the resulting integrand.

Rule: If  $a^2 - b^2 \neq 0 \wedge a+b \neq 0$ , then

$$\int \sqrt{a+b \sin[c+dx]} \, dx \rightarrow \frac{\sqrt{a+b \sin[c+dx]}}{\sqrt{\frac{a+b \sin[c+dx]}{a+b}}} \int \sqrt{\frac{a}{a+b} + \frac{b}{a+b} \sin[c+dx]} \, dx$$

Program code:

```
Int[Sqrt[a+_.*sin[c_+d_.*x_]],x_Symbol] :=
  Sqrt[a+b*sin[c+d*x]]/Sqrt[(a+b*sin[c+d*x])/(a+b)]*Int[Sqrt[a/(a+b)+b/(a+b)*Sin[c+d*x]],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && Not[GtQ[a+b,0]]
```

**2:**  $\int (a+b \sin[c+dx])^n \, dx$  when  $a^2 - b^2 \neq 0 \wedge n > 1 \wedge 2n \in \mathbb{Z}$

Derivation: Nondegenerate sine recurrence 1b with  $A \rightarrow a c$ ,  $B \rightarrow b c + a d$ ,  $C \rightarrow b d$ ,  $m \rightarrow -1 + m$ ,  $n \rightarrow -1$ ,  $p \rightarrow 0$

Rule: If  $a^2 - b^2 \neq 0 \wedge n > 1 \wedge 2n \in \mathbb{Z}$ , then

$$\int (a+b \sin[c+dx])^n \, dx \rightarrow -\frac{b \cos[c+dx] (a+b \sin[c+dx])^{n-1}}{d n} + \frac{1}{n} \int (a+b \sin[c+dx])^{n-2} (a^2 n + b^2 (n-1) + a b (2n-1) \sin[c+dx]) \, dx$$

Program code:

```
Int[(a+_.*sin[c_+d_.*x_])^n_,x_Symbol] :=
  -b*cos[c+d*x]*(a+b*sin[c+d*x])^(n-1)/(d*n) +
  1/n*Int[(a+b*sin[c+d*x])^(n-2)*Simp[a^2*n+b^2*(n-1)+a*b*(2*n-1)*Sin[c+d*x],x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && GtQ[n,1] && IntegerQ[2*n]
```

**2.**  $\int (a+b \sin[c+dx])^n \, dx$  when  $a^2 - b^2 \neq 0 \wedge 2n \in \mathbb{Z}^-$

$$1. \int \frac{1}{a + b \sin[c + dx]} dx \text{ when } a^2 - b^2 \neq 0$$

$$1. \int \frac{1}{a + b \sin[c + dx]} dx \text{ when } a^2 - b^2 > 0$$

$$\textcolor{red}{1}: \int \frac{1}{a + b \sin[c + dx]} dx \text{ when } a^2 - b^2 > 0 \wedge a > 0$$

Note: Resulting antiderivative is continuous on the real line.

Rule: If  $a^2 - b^2 > 0 \wedge a > 0$ , let  $q = \sqrt{a^2 - b^2}$ , then

$$\int \frac{1}{a + b \sin[c + dx]} dx \rightarrow \frac{x}{q} + \frac{2}{dq} \operatorname{ArcTan}\left[\frac{b \cos[c + dx]}{a + q + b \sin[c + dx]}\right]$$

Program code:

```
Int[1/(a+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  With[{q=Rt[a^2-b^2,2]},
    x/q + 2/(d*q)*ArcTan[b*cos[c+d*x]/(a+q+b*sin[c+d*x])]] /;
FreeQ[{a,b,c,d},x] && GtQ[a^2-b^2,0] && PosQ[a]
```

**2:**  $\int \frac{1}{a + b \sin[c + dx]} dx$  when  $a^2 - b^2 > 0 \wedge a \neq 0$

Note: Resulting antiderivative is continuous on the real line.

Rule: If  $a^2 - b^2 > 0 \wedge a \neq 0$ , let  $q = \sqrt{a^2 - b^2}$ , then

$$\int \frac{1}{a + b \sin[c + dx]} dx \rightarrow -\frac{x}{q} - \frac{2}{dq} \operatorname{ArcTan}\left[\frac{b \cos[c + dx]}{a - q + b \sin[c + dx]}\right]$$

Program code:

```
Int[1/(a_+b_.*sin[c_+d_.*x_]),x_Symbol] :=
  With[{q=Rt[a^2-b^2,2]},
    -x/q - 2/(d*q)*ArcTan[b*cos[c+d*x]/(a-q+b*sin[c+d*x])]] /;
  FreeQ[{a,b,c,d},x] && GtQ[a^2-b^2,0] && NegQ[a]
```



$$\text{2: } \int \frac{1}{a+b \sin[c+dx]} dx \text{ when } a^2 - b^2 \neq 0$$

Reference: G&R 2.551.3, CRC 340, A&S 4.3.131

Reference: G&R 2.553.3, CRC 341, A&S 4.3.133

Derivation: Integration by substitution

Basis:  $F[\sin[c+dx], \cos[c+dx]] =$

$$\frac{2}{d} \text{Subst}\left[\frac{1}{1+x^2} F\left[\frac{2x}{1+x^2}, \frac{1-x^2}{1+x^2}\right], x, \tan\left[\frac{1}{2}(c+dx)\right]\right] \partial_x \tan\left[\frac{1}{2}(c+dx)\right]$$

$$\text{Basis: } \frac{1}{a+b \sin[c+dx]} = \frac{2}{d} \text{Subst}\left[\frac{1}{a+2bx+ax^2}, x, \tan\left[\frac{1}{2}(c+dx)\right]\right] \partial_x \tan\left[\frac{1}{2}(c+dx)\right]$$

$$\text{Basis: } \frac{1}{a+b \cos[c+dx]} = \frac{2}{d} \text{Subst}\left[\frac{1}{a+b+(a-b)x^2}, x, \tan\left[\frac{1}{2}(c+dx)\right]\right] \partial_x \tan\left[\frac{1}{2}(c+dx)\right]$$

$$\text{Note: } \tan\left[\frac{z}{2}\right] = \frac{\sin[z]}{1+\cos[z]}$$

Rule: If  $a^2 - b^2 \neq 0$ , then

$$\int \frac{1}{a+b \sin[c+dx]} dx \rightarrow \frac{2}{d} \text{Subst}\left[\int \frac{1}{a+2bx+ax^2} dx, x, \tan\left[\frac{1}{2}(c+dx)\right]\right]$$

$$\int \frac{1}{a+b \cos[c+dx]} dx \rightarrow \frac{2}{d} \text{Subst}\left[\int \frac{1}{a+b+(a-b)x^2} dx, x, \tan\left[\frac{1}{2}(c+dx)\right]\right]$$

Program code:

```
Int[1/(a+_b_.sin[c_+Pi/2+d_.*x_]),x_Symbol] :=
  With[{e=FreeFactors[Tan[(c+d*x)/2],x]},
    2*e/d*Subst[Int[1/(a+b+(a-b)*e^2*x^2),x],x,Tan[(c+d*x)/2]/e]] /;
  FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0]
```

```

Int[1/(a+b_.*sin[c_.+d_.*x_]),x_Symbol] :=
  With[{e=FreeFactors[Tan[(c+d*x)/2],x]},
    2*e/d*Subst[Int[1/(a+2*b*e*x+a*e^2*x^2),x],x,Tan[(c+d*x)/2]/e]] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0]

```

$$2. \int \frac{1}{\sqrt{a+b \sin[c+dx]}} dx \text{ when } a^2 - b^2 \neq 0$$

$$1: \int \frac{1}{\sqrt{a+b \sin[c+dx]}} dx \text{ when } a^2 - b^2 \neq 0 \wedge a+b > 0$$

Derivation: Primitive rule

Basis: If  $a+b > 0$ , then  $\partial_x \text{EllipticF}\left[\frac{1}{2}\left(x - \frac{\pi}{2}\right), \frac{2b}{a+b}\right] = \frac{\sqrt{a+b}}{2\sqrt{a+b \sin[x]}}$

Rule: If  $a^2 - b^2 \neq 0 \wedge a+b > 0$ , then

$$\int \frac{1}{\sqrt{a+b \sin[c+dx]}} dx \rightarrow \frac{2}{d\sqrt{a+b}} \text{EllipticF}\left[\frac{1}{2}\left(c - \frac{\pi}{2} + dx\right), \frac{2b}{a+b}\right]$$

Program code:

```

Int[1/Sqrt[a+b_.*sin[c_.+d_.*x_]],x_Symbol] :=
  2/(d*Sqrt[a+b])*EllipticF[1/2*(c-Pi/2+d*x),2*b/(a+b)] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && GtQ[a+b,0]

```

$$2: \int \frac{1}{\sqrt{a+b \sin[c+dx]}} dx \text{ when } a^2 - b^2 \neq 0 \wedge a-b > 0$$

Derivation: Primitive rule

Basis: If  $a - b > 0$ , then  $\partial_x \text{EllipticF}\left[\frac{1}{2}\left(x + \frac{\pi}{2}\right), -\frac{2b}{a-b}\right] = \frac{\sqrt{a-b}}{2\sqrt{a+b \sin[x]}}$

Rule: If  $a^2 - b^2 \neq 0 \wedge a - b > 0$ , then

$$\int \frac{1}{\sqrt{a+b \sin[c+dx]}} dx \rightarrow \frac{2}{d\sqrt{a-b}} \text{EllipticF}\left[\frac{1}{2}\left(c + \frac{\pi}{2} + dx\right), -\frac{2b}{a-b}\right]$$

Program code:

```
Int[1/Sqrt[a_+b_.*sin[c_+d_.*x_]],x_Symbol] :=
  2/(d*Sqrt[a-b])*EllipticF[1/2*(c+Pi/2+d*x),-2*b/(a-b)] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && GtQ[a-b,0]
```

$$\text{3: } \int \frac{1}{\sqrt{a+b \sin[c+dx]}} dx \text{ when } a^2 - b^2 \neq 0 \wedge a+b \neq 0$$

Derivation: Piecewise constant extraction

$$\text{Basis: } \partial_x \frac{\sqrt{\frac{a+b \sin[c+dx]}{a+b}}}{\sqrt{a+b \sin[c+dx]}} == 0$$

Note: Since  $\frac{a}{a+b} + \frac{b}{a+b} \sin[c+dx] == 1 > 0$ , rule f1 applies to the resulting integrand.

Rule: If  $a^2 - b^2 \neq 0 \wedge a+b \neq 0$ , then

$$\int \frac{1}{\sqrt{a+b \sin[c+dx]}} dx \rightarrow \frac{\sqrt{\frac{a+b \sin[c+dx]}{a+b}}}{\sqrt{a+b \sin[c+dx]}} \int \frac{1}{\sqrt{\frac{a}{a+b} + \frac{b}{a+b} \sin[c+dx]}} dx$$

Program code:

```
Int[1/Sqrt[a_+b_.*sin[c_+d_.x_]],x_Symbol] :=
  Sqrt[(a+b*sin[c+d*x])/(a+b)]/Sqrt[a+b*sin[c+d*x]]*Int[1/Sqrt[a/(a+b)+b/(a+b)*sin[c+d*x]],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && Not[GtQ[a+b,0]]
```

$$\mathbf{3:} \int (a + b \sin[c + dx])^n dx \text{ when } a^2 - b^2 \neq 0 \wedge n < -1 \wedge 2n \in \mathbb{Z}$$

Reference: G&R 2.552.3

Derivation: Nondegenerate sine recurrence 1a with  $A \rightarrow 1$ ,  $B \rightarrow 0$ ,  $C \rightarrow 0$ ,  $m \rightarrow 0$ ,  $p \rightarrow 0$

Rule: If  $a^2 - b^2 \neq 0 \wedge n < -1 \wedge 2n \in \mathbb{Z}$ , then

$$\int (a + b \sin[c + dx])^n dx \rightarrow -\frac{b \cos[c + dx] (a + b \sin[c + dx])^{n+1}}{d(n+1)(a^2 - b^2)} + \frac{1}{(n+1)(a^2 - b^2)} \int (a + b \sin[c + dx])^{n+1} (a(n+1) - b(n+2) \sin[c + dx]) dx$$

Program code:

```
Int[(a_+b_.sin[c_+d_.x_])^n_,x_Symbol] :=
  -b*cos[c+d*x]*(a+b*sin[c+d*x])^(n+1)/(d*(n+1)*(a^2-b^2)) +
  1/((n+1)*(a^2-b^2))*Int[(a+b*sin[c+d*x])^(n+1)*Simp[a*(n+1)-b*(n+2)*sin[c+d*x],x],x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && LtQ[n,-1] && IntegerQ[2*n]
```

$$\mathbf{2:} \int (a + b \sin[c + dx])^n dx \text{ when } a^2 - b^2 \neq 0 \wedge 2n \notin \mathbb{Z}$$

Derivation: Piecewise constant extraction and integration by substitution

$$\text{Basis: } \partial_x \frac{\cos[c+dx]}{\sqrt{1+\sin[c+dx]} \sqrt{1-\sin[c+dx]}} = 0$$

$$\text{Basis: } \cos[c + dx] = \frac{1}{d} \partial_x \sin[c + dx]$$

Rule: If  $a^2 - b^2 \neq 0 \wedge 2n \notin \mathbb{Z}$ , then

$$\int (a + b \sin[c + dx])^n dx \rightarrow$$

$$\frac{\cos[c + dx]}{\sqrt{1 + \sin[c + dx]} \sqrt{1 - \sin[c + dx]}} \int \frac{\cos[c + dx] (a + b \sin[c + dx])^n}{\sqrt{1 + \sin[c + dx]} \sqrt{1 - \sin[c + dx]}} dx \rightarrow$$

$$\frac{\cos[c + dx]}{d \sqrt{1 + \sin[c + dx]} \sqrt{1 - \sin[c + dx]}} \text{Subst} \left[ \int \frac{(a + bx)^n}{\sqrt{1+x} \sqrt{1-x}} dx, x, \sin[c + dx] \right]$$

Program code:

```
Int[(a_+b_.sin[c_+d_.x_])^n_,x_Symbol] :=
  Cos[c+d*x]/(d*Sqrt[1+Sin[c+d*x]]*Sqrt[1-Sin[c+d*x]])*Subst[Int[(a+b*x)^n/(Sqrt[1+x]*Sqrt[1-x]),x],x,Sin[c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && NeQ[a^2-b^2,0] && Not[IntegerQ[2*n]]
```

## Rules for integrands of the form $(a + b \sin[c + dx] \cos[c + dx])^n$

**1:**  $\int (a + b \sin[c + dx] \cos[c + dx])^n dx$

Derivation: Algebraic simplification

Basis:  $\sin[z] \cos[z] == \frac{1}{2} \sin[2z]$

Rule:

$$\int (a + b \sin[c + dx] \cos[c + dx])^n dx \rightarrow \int \left( a + \frac{1}{2} b \sin[2c + 2dx] \right)^n dx$$

Program code:

```
Int[(a+b_.*sin[c_.+d_.*x_] *cos[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[(a+b*Sin[2*c+2*d*x]/2)^n,x] /;
FreeQ[{a,b,c,d,n},x]
```