Rules for integrands of the form $(d\,\mathrm{Tan}[e + f\,x])^n\,(a + b\,\mathrm{Sec}[e + f\,x])^m$

1. $\int \mathrm{Tan}[c + d\,x]^m\,(a + b\,\mathrm{Sec}[c + d\,x])^n\,dx$ when $\frac{m-1}{2} \in \mathbb{Z} \land a^2 - b^2 == 0$

   1: $\int \mathrm{Tan}[c + d\,x]^m\,(a + b\,\mathrm{Sec}[c + d\,x])^n\,dx$ when $\frac{m-1}{2} \in \mathbb{Z} \land a^2 - b^2 == 0 \land n \in \mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m-1}{2} \in \mathbb{Z} \land a^2 - b^2 == 0 \land n \in \mathbb{Z}$, then

$$\mathrm{Tan}[c + d\,x]^m\,(a + b\,\mathrm{Sec}[c + d\,x])^n == -\frac{1}{a^{m-n-1}\,b^n\,d}\,\mathrm{Subst}\left[\frac{(a-b\,x)^{\frac{m-1}{2}}\,(a+b\,x)^{\frac{m-1}{2}+n}}{x^{m+n}},\,x,\,\mathrm{Cos}[c + d\,x]\right]\,\partial_x\,\mathrm{Cos}[c + d\,x]$$

Rule: If $\frac{m-1}{2} \in \mathbb{Z} \land a^2 - b^2 == 0 \land n \in \mathbb{Z}$, then

$$\int \mathrm{Tan}[c + d\,x]^m\,(a + b\,\mathrm{Sec}[c + d\,x])^n\,dx \longrightarrow -\frac{1}{a^{m-n-1}\,b^n\,d}\,\mathrm{Subst}\left[\int \frac{(a - b\,x)^{\frac{m-1}{2}}\,(a + b\,x)^{\frac{m-1}{2}+n}}{x^{m+n}}\,dx,\,x,\,\mathrm{Cos}[c + d\,x]\right]$$

Program code:

```
Int[cot[c_.+d_.*x_]^m_.*(a_+b_.*csc[c_.+d_.*x_])^n_.,x_Symbol] :=
  1/(a^(m-n-1)*b^n*d)*Subst[Int[(a-b*x)^((m-1)/2)*(a+b*x)^((m-1)/2+n)/x^(m+n),x],x,Sin[c+d*x]] /;
FreeQ[{a,b,c,d},x] && IntegerQ[(m-1)/2] && EqQ[a^2-b^2,0] && IntegerQ[n]
```

2: $\int \text{Tan}[c+d\,x]^m\,(a+b\,\text{Sec}[c+d\,x])^n\,dx$ when $\frac{m+1}{2}\in\mathbb{Z}\,\land\,a^2-b^2==0\,\land\,n\notin\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $\frac{m-1}{2}\in\mathbb{Z}\,\land\,a^2-b^2==0$, then

$$\text{Tan}[c+d\,x]^m == \frac{1}{d\,b^{m-1}}\,\text{Subst}\Big[\frac{(-a+b\,x)^{\frac{m-1}{2}}\,(a+b\,x)^{\frac{m-1}{2}}}{x},\,x,\,\text{Sec}[c+d\,x]\Big]\,\partial_x\,\text{Sec}[c+d\,x]$$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}\,\land\,a^2-b^2==0$, then

$$\int \text{Tan}[c+d\,x]^m\,(a+b\,\text{Sec}[c+d\,x])^n\,dx \,\rightarrow\, \frac{1}{d\,b^{m-1}}\,\text{Subst}\Big[\int\frac{(-a+b\,x)^{\frac{m-1}{2}}\,(a+b\,x)^{\frac{m-1}{2}+n}}{x}\,dx,\,x,\,\text{Sec}[c+d\,x]\Big]$$

Program code:

```
Int[cot[c_.+d_.*x_]^m_.*(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  -1/(d*b^(m-1))*Subst[Int[(-a+b*x)^((m-1)/2)*(a+b*x)^((m-1)/2+n)/x,x],x,Csc[c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && IntegerQ[(m-1)/2] && EqQ[a^2-b^2,0] && Not[IntegerQ[n]]
```

2. $\int (e \tan[c+d x])^m (a+b \sec[c+d x]) \, dx$

1: $\int (e \tan[c+d x])^m (a+b \sec[c+d x]) \, dx$ when $m>1$

Rule: If $m>1$, then

$$\int (e \tan[c+d x])^m (a+b \sec[c+d x]) \, dx \to$$
$$\frac{e (e \tan[c+d x])^{m-1} (a m+b (m-1) \sec[c+d x])}{d m (m-1)}-\frac{e^2}{m} \int (e \tan[c+d x])^{m-2} (a m+b (m-1) \sec[c+d x]) \, dx$$

Program code:

```
Int[(e_.*cot[c_.+d_.*x_])^m_*(a_+b_.*csc[c_.+d_.*x_]),x_Symbol] :=
  -e*(e*Cot[c+d*x])^(m-1)*(a*m+b*(m-1)*Csc[c+d*x])/(d*m*(m-1)) -
  e^2/m*Int[(e*Cot[c+d*x])^(m-2)*(a*m+b*(m-1)*Csc[c+d*x]),x] /;
FreeQ[{a,b,c,d,e},x] && GtQ[m,1]
```

2: $\int (e\,\mathrm{Tan}[c+d\,x])^m\,(a+b\,\mathrm{Sec}[c+d\,x])\,dx$ when $m<-1$

Rule: If $m<-1$, then

$$\int (e\,\mathrm{Tan}[c+d\,x])^m\,(a+b\,\mathrm{Sec}[c+d\,x])\,dx\;\longrightarrow$$
$$\frac{(e\,\mathrm{Tan}[c+d\,x])^{m+1}\,(a+b\,\mathrm{Sec}[c+d\,x])}{d\,e\,(m+1)}-\frac{1}{e^2\,(m+1)}\int (e\,\mathrm{Tan}[c+d\,x])^{m+2}\,(a\,(m+1)+b\,(m+2)\,\mathrm{Sec}[c+d\,x])\,dx$$

Program code:

```
Int[(e_.*cot[c_.+d_.*x_])^m_*(a_+b_.*csc[c_.+d_.*x_]),x_Symbol] :=
  -(e*Cot[c+d*x])^(m+1)*(a+b*Csc[c+d*x])/(d*e*(m+1)) -
  1/(e^2*(m+1))*Int[(e*Cot[c+d*x])^(m+2)*(a*(m+1)+b*(m+2)*Csc[c+d*x]),x] /;
FreeQ[{a,b,c,d,e},x] && LtQ[m,-1]
```

3: $\int \frac{a + b \, \text{Sec}[c + d \, x]}{\text{Tan}[c + d \, x]} \, dx$

Derivation: Algebraic simplification

Basis: $\frac{a+b \, \text{Sec}[z]}{\text{Tan}[z]} == \frac{b+a \, \text{Cos}[z]}{\text{Sin}[z]}$

Rule:

$$\int \frac{a + b \, \text{Sec}[c + d \, x]}{\text{Tan}[c + d \, x]} \, dx \longrightarrow \int \frac{b + a \, \text{Cos}[c + d \, x]}{\text{Sin}[c + d \, x]} \, dx$$

Program code:

```
Int[(a_+b_.*csc[c_.+d_.*x_])/cot[c_.+d_.*x_],x_Symbol] :=
  Int[(b+a*Sin[c+d*x])/Cos[c+d*x],x] /;
FreeQ[{a,b,c,d},x]
```

4: $\int (e \operatorname{Tan}[c+d x])^m (a+b \operatorname{Sec}[c+d x]) \, dx$

Derivation: Algebraic expansion

Rule:

$$\int (e \operatorname{Tan}[c+d x])^m (a+b \operatorname{Sec}[c+d x]) \, dx \longrightarrow a \int (e \operatorname{Tan}[c+d x])^m \, dx + b \int (e \operatorname{Tan}[c+d x])^m \operatorname{Sec}[c+d x] \, dx$$

Program code:

```
Int[(e_.*cot[c_.+d_.*x_])^m_.*(a_+b_.*csc[c_.+d_.*x_]),x_Symbol] :=
  a*Int[(e*Cot[c+d*x])^m,x] + b*Int[(e*Cot[c+d*x])^m*Csc[c+d*x],x] /;
FreeQ[{a,b,c,d,e,m},x]
```

3: $\int \operatorname{Tan}[c+d\,x]^m\,(a+b\operatorname{Sec}[c+d\,x])^n\,dx$ when $\frac{m-1}{2}\in\mathbb{Z}\,\wedge\,a^2-b^2\neq 0$

Derivation: Integration by substitution

Basis: If $\frac{m-1}{2}\in\mathbb{Z}$, then $\operatorname{Tan}[c+d\,x]^m == \frac{(-1)^{\frac{m-1}{2}}}{d\,b^{m-1}}\operatorname{Subst}\left[\frac{\left(b^2-x^2\right)^{\frac{m-1}{2}}}{x},\,x,\,b\operatorname{Sec}[c+d\,x]\right]\partial_x\,(b\operatorname{Sec}[c+d\,x])$

Rule: If $\frac{m-1}{2}\in\mathbb{Z}\,\wedge\,a^2-b^2\neq 0$, then

$$\int \operatorname{Tan}[c+d\,x]^m\,(a+b\operatorname{Sec}[c+d\,x])^n\,dx \longrightarrow \frac{(-1)^{\frac{m-1}{2}}}{d\,b^{m-1}}\operatorname{Subst}\left[\int\frac{\left(b^2-x^2\right)^{\frac{m-1}{2}}(a+x)^n}{x}\,dx,\,x,\,b\operatorname{Sec}[c+d\,x]\right]$$

Program code:

```
Int[cot[c_.+d_.*x_]^m_.*(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  -(-1)^((m-1)/2)/(d*b^(m-1))*Subst[Int[(b^2-x^2)^((m-1)/2)*(a+x)^n/x,x],x,b*Csc[c+d*x]] /;
FreeQ[{a,b,c,d,n},x] && IntegerQ[(m-1)/2] && NeQ[a^2-b^2,0]
```

4: $\int (e \tan[c+d x])^m (a+b \sec[c+d x])^n \, dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n \in \mathbb{Z}^+$, then

$$\int (e \tan[c+d x])^m (a+b \sec[c+d x])^n \, dx \longrightarrow \int (e \tan[c+d x])^m \operatorname{ExpandIntegrand}\left[(a+b \sec[c+d x])^n, x\right] dx$$

Program code:

```
Int[(e_.*cot[c_.+d_.*x_])^m_*(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(e*Cot[c+d*x])^m,(a+b*Csc[c+d*x])^n,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && IGtQ[n,0]
```

5. $\int (e\,\text{Tan}[c+d\,x])^m\,(a+b\,\text{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2==0$

1: $\int \text{Tan}[c+d\,x]^m\,(a+b\,\text{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2==0\ \land\ \frac{m}{2}\in\mathbb{Z}\ \land\ n-\frac{1}{2}\in\mathbb{Z}$

Derivation: Integration by substitution

Basis: If $a^2-b^2==0\ \land\ \frac{m}{2}\in\mathbb{Z}\ \land\ n-\frac{1}{2}\in\mathbb{Z}$, then

$\text{Tan}[c+d\,x]^m\,(a+b\,\text{Sec}[c+d\,x])^n ==$

$$\frac{2\,a^{\frac{m}{2}+n+\frac{1}{2}}}{d}\,\text{Subst}\left[\frac{x^m\,(2+a\,x^2)^{\frac{m}{2}+n-\frac{1}{2}}}{(1+a\,x^2)},\ x,\ \frac{\text{Tan}[c+d\,x]}{\sqrt{a+b\,\text{Sec}[c+d\,x]}}\right]\partial_x\frac{\text{Tan}[c+d\,x]}{\sqrt{a+b\,\text{Sec}[c+d\,x]}}$$

Rule: If $a^2-b^2==0\ \land\ \frac{m}{2}\in\mathbb{Z}\ \land\ n-\frac{1}{2}\in\mathbb{Z}$, then

$$\int \text{Tan}[c+d\,x]^m\,(a+b\,\text{Sec}[c+d\,x])^n\,dx\ \longrightarrow\ \frac{2\,a^{\frac{m}{2}+n+\frac{1}{2}}}{d}\,\text{Subst}\left[\int\frac{x^m\,(2+a\,x^2)^{\frac{m}{2}+n-\frac{1}{2}}}{(1+a\,x^2)}\,dx,\ x,\ \frac{\text{Tan}[c+d\,x]}{\sqrt{a+b\,\text{Sec}[c+d\,x]}}\right]$$

Program code:

```
Int[cot[c_.+d_.*x_]^m_.*(a_+b_.*csc[c_.+d_.*x_])^n_.,x_Symbol] :=
  -2*a^(m/2+n+1/2)/d*Subst[Int[x^m*(2+a*x^2)^(m/2+n-1/2)/(1+a*x^2),x],x,Cot[c+d*x]/Sqrt[a+b*Csc[c+d*x]]] /;
FreeQ[{a,b,c,d},x] && EqQ[a^2-b^2,0] && IntegerQ[m/2] && IntegerQ[n-1/2]
```

2: $\int (e \operatorname{Tan}[c+d x])^m (a+b \operatorname{Sec}[c+d x])^n \, dx$ when $a^2-b^2=0 \land n \in \mathbb{Z}^-$

Derivation: Algebraic simplification

Basis: If $a^2-b^2=0$, then $a+b \operatorname{Sec}[z]=a^2 e^{-2} (e \operatorname{Tan}[z])^2 (-a+b \operatorname{Sec}[z])^{-1}$

Rule: If $a^2-b^2=0 \land n \in \mathbb{Z}^-$, then

$$\int (e \operatorname{Tan}[c+d x])^m (a+b \operatorname{Sec}[c+d x])^n \, dx \rightarrow a^{2 n} e^{-2 n} \int (e \operatorname{Tan}[c+d x])^{m+2 n} (-a+b \operatorname{Sec}[c+d x])^{-n} \, dx$$

Program code:

```
Int[(e_.*cot[c_.+d_.*x_])^m_*(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  a^(2*n)*e^(-2*n)*Int[(e*Cot[c+d*x])^(m+2*n)/(-a+b*Csc[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,e,m},x] && EqQ[a^2-b^2,0] && ILtQ[n,0]
```

3: $\int (e \operatorname{Tan}[c+d x])^m (a+b \operatorname{Sec}[c+d x])^n \, dx$ when $a^2-b^2==0 \land n \notin \mathbb{Z}$

Rule: If $a^2-b^2==0$, then

$$\int (e \operatorname{Tan}[c+d x])^m (a+b \operatorname{Sec}[c+d x])^n \, dx \longrightarrow$$

$$\frac{2^{m+n+1} (e \operatorname{Tan}[c+d x])^{m+1} (a+b \operatorname{Sec}[c+d x])^n}{d e (m+1)} \left(\frac{a}{a+b \operatorname{Sec}[c+d x]}\right)^{m+n+1} \operatorname{AppellF1}\left[\frac{m+1}{2}, m+n, 1, \frac{m+3}{2}, -\frac{a-b \operatorname{Sec}[c+d x]}{a+b \operatorname{Sec}[c+d x]}, \frac{a-b \operatorname{Sec}[c+d x]}{a+b \operatorname{Sec}[c+d x]}\right]$$

Program code:

```
Int[(e_.*cot[c_.+d_.*x_])^m_*(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  -2^(m+n+1)*(e*Cot[c+d*x])^(m+1)*(a+b*Csc[c+d*x])^n/(d*e*(m+1))*(a/(a+b*Csc[c+d*x]))^(m+n+1)*
    AppellF1[(m+1)/2,m+n,1,(m+3)/2,-(a-b*Csc[c+d*x])/(a+b*Csc[c+d*x]),(a-b*Csc[c+d*x])/(a+b*Csc[c+d*x])] /;
FreeQ[{a,b,c,d,e,m,n},x] && EqQ[a^2-b^2,0] && Not[IntegerQ[n]]
```

6. $\int (e \operatorname{Tan}[c+d x])^m (a+b \operatorname{Sec}[c+d x])^n \, dx$ when $a^2-b^2 \neq 0$

1. $\int \frac{(e \operatorname{Tan}[c+d x])^m}{a+b \operatorname{Sec}[c+d x]} \, dx$ when $a^2-b^2 \neq 0 \land m-\frac{1}{2} \in \mathbb{Z}$

1. $\int \frac{(e \operatorname{Tan}[c+d x])^m}{a+b \operatorname{Sec}[c+d x]} \, dx$ when $a^2-b^2 \neq 0 \land m+\frac{1}{2} \in \mathbb{Z}^+$

1: $\int \frac{\sqrt{e \operatorname{Tan}[c+d x]}}{a+b \operatorname{Sec}[c+d x]} \, dx$ when $a^2-b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b \operatorname{Sec}[z]} == \frac{1}{a} - \frac{b}{a (b+a \operatorname{Cos}[z])}$

Rule: If $a^2-b^2 \neq 0$, then

$$\int \frac{\sqrt{e\,\mathrm{Tan}[c+d\,x]}}{a+b\,\mathrm{Sec}[c+d\,x]}\,dx \longrightarrow \frac{1}{a}\int \sqrt{e\,\mathrm{Tan}[c+d\,x]}\,dx - \frac{b}{a}\int \frac{\sqrt{e\,\mathrm{Tan}[c+d\,x]}}{b+a\,\mathrm{Cos}[c+d\,x]}\,dx$$

Program code:

```
Int[Sqrt[e_.*cot[c_.+d_.*x_]]/(a_+b_.*csc[c_.+d_.*x_]),x_Symbol] :=
  1/a*Int[Sqrt[e*Cot[c+d*x]],x] - b/a*Int[Sqrt[e*Cot[c+d*x]]/(b+a*Sin[c+d*x]),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[a^2-b^2,0]
```

2: $\int \frac{(e\,\mathrm{Tan}[c+d\,x])^m}{a+b\,\mathrm{Sec}[c+d\,x]}\,dx$ when $a^2-b^2\neq 0 \land m-\frac{1}{2}\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\frac{\mathrm{Tan}[z]^2}{a+b\,\mathrm{Sec}[z]} == -\frac{a-b\,\mathrm{Sec}[z]}{b^2}+\frac{a^2-b^2}{b^2\,(a+b\,\mathrm{Sec}[z])}$

Rule: If $a^2-b^2\neq 0 \land m-\frac{1}{2}\in\mathbb{Z}^+$, then

$$\int \frac{(e\,\mathrm{Tan}[c+d\,x])^m}{a+b\,\mathrm{Sec}[c+d\,x]}\,dx \longrightarrow -\frac{e^2}{b^2}\int (e\,\mathrm{Tan}[c+d\,x])^{m-2}\,(a-b\,\mathrm{Sec}[c+d\,x])\,dx+\frac{e^2\,(a^2-b^2)}{b^2}\int \frac{(e\,\mathrm{Tan}[c+d\,x])^{m-2}}{a+b\,\mathrm{Sec}[c+d\,x]}\,dx$$

Program code:

```
Int[(e_.*cot[c_.+d_.*x_])^m_/(a_+b_.*csc[c_.+d_.*x_]),x_Symbol] :=
  -e^2/b^2*Int[(e*Cot[c+d*x])^(m-2)*(a-b*Csc[c+d*x]),x] +
  e^2*(a^2-b^2)/b^2*Int[(e*Cot[c+d*x])^(m-2)/(a+b*Csc[c+d*x]),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[a^2-b^2,0] && IGtQ[m-1/2,0]
```

2. $\int \frac{(e\,\mathrm{Tan}[c+d\,x])^m}{a+b\,\mathrm{Sec}[c+d\,x]}\,dx$ when $a^2-b^2\neq 0 \land m-\frac{1}{2}\in\mathbb{Z}^-$

1: $\int \frac{1}{\sqrt{e \operatorname{Tan}[c+d x]}\,(a+b \operatorname{Sec}[c+d x])}\,dx$ when $a^2-b^2 \neq 0$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b\operatorname{Sec}[z]} == \frac{1}{a} - \frac{b}{a\,(b+a\operatorname{Cos}[z])}$

Rule: If $a^2-b^2 \neq 0$, then

$$\int \frac{1}{\sqrt{e \operatorname{Tan}[c+d x]}\,(a+b \operatorname{Sec}[c+d x])}\,dx \to \frac{1}{a}\int \frac{1}{\sqrt{e \operatorname{Tan}[c+d x]}}\,dx - \frac{b}{a}\int \frac{1}{\sqrt{e \operatorname{Tan}[c+d x]}\,(b+a \operatorname{Cos}[c+d x])}\,dx$$

Program code:

```
Int[1/(Sqrt[e_.*cot[c_.+d_.*x_]]*(a_+b_.*csc[c_.+d_.*x_])),x_Symbol] :=
  1/a*Int[1/Sqrt[e*Cot[c+d*x]],x] - b/a*Int[1/(Sqrt[e*Cot[c+d*x]]*(b+a*Sin[c+d*x])),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[a^2-b^2,0]
```

2: $\int \frac{(e \operatorname{Tan}[c+d x])^m}{a+b \operatorname{Sec}[c+d x]} \, dx$ when $a^2-b^2 \neq 0 \land m+\frac{1}{2} \in \mathbb{Z}^-$

Derivation: Algebraic expansion

Basis: $\frac{1}{a+b \operatorname{Sec}[z]} == \frac{a-b \operatorname{Sec}[z]}{a^2-b^2} + \frac{b^2 \operatorname{Tan}[z]^2}{(a^2-b^2)(a+b \operatorname{Sec}[z])}$

Rule: If $a^2-b^2 \neq 0 \land m+\frac{1}{2} \in \mathbb{Z}^-$, then

$$\int \frac{(e \operatorname{Tan}[c+d x])^m}{a+b \operatorname{Sec}[c+d x]} \, dx \to \frac{1}{a^2-b^2} \int (e \operatorname{Tan}[c+d x])^m (a-b \operatorname{Sec}[c+d x]) \, dx + \frac{b^2}{e^2 (a^2-b^2)} \int \frac{(e \operatorname{Tan}[c+d x])^{m+2}}{a+b \operatorname{Sec}[c+d x]} \, dx$$

Program code:

```
Int[(e_.*cot[c_.+d_.*x_])^m_/(a_+b_.*csc[c_.+d_.*x_]),x_Symbol] :=
  1/(a^2-b^2)*Int[(e*Cot[c+d*x])^m*(a-b*Csc[c+d*x]),x] +
  b^2/(e^2*(a^2-b^2))*Int[(e*Cot[c+d*x])^(m+2)/(a+b*Csc[c+d*x]),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[a^2-b^2,0] && ILtQ[m+1/2,0]
```

2. $\int \text{Tan}[c+d\,x]^m\,(a+b\,\text{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2\neq 0\ \land\ \frac{m}{2}\in\mathbb{Z}$

1. $\int \text{Tan}[c+d\,x]^m\,(a+b\,\text{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2\neq 0\ \land\ \frac{m}{2}\in\mathbb{Z}^+$

1: $\int \text{Tan}[c+d\,x]^2\,(a+b\,\text{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2\neq 0$

Derivation: Algebraic expansion

Basis: $\text{Tan}[z]^2 == -1+\text{Sec}[z]^2$

Rule: If $a^2-b^2\neq 0$, then

$$\int \text{Tan}[c+d\,x]^2\,(a+b\,\text{Sec}[c+d\,x])^n\,dx\ \longrightarrow\ \int \left(-1+\text{Sec}[c+d\,x]^2\right)\,(a+b\,\text{Sec}[c+d\,x])^n\,dx$$

Program code:

```
Int[cot[c_.+d_.*x_]^2*(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[(-1+Csc[c+d*x]^2)*(a+b*Csc[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[a^2-b^2,0]
```

2: $\int \text{Tan}[c+d\,x]^m\,(a+b\,\text{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2\neq 0\,\land\,\frac{m}{2}\in\mathbb{Z}^+\land\,n-\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\text{Tan}[z]^2 == -1+\text{Sec}[z]^2$

Rule: If $a^2-b^2\neq 0\,\land\,\frac{m}{2}\in\mathbb{Z}^+\land\,n-\frac{1}{2}\in\mathbb{Z}$, then

$$\int \text{Tan}[c+d\,x]^m\,(a+b\,\text{Sec}[c+d\,x])^n\,dx\,\longrightarrow\,\int (a+b\,\text{Sec}[c+d\,x])^n\,\text{ExpandIntegrand}\left[\left(-1+\text{Sec}[c+d\,x]^2\right)^{m/2},\,x\right]dx$$

Program code:

```
Int[cot[c_.+d_.*x_]^m_*(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*Csc[c+d*x])^n,(-1+Csc[c+d*x]^2)^(m/2),x],x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[a^2-b^2,0] && IGtQ[m/2,0] && IntegerQ[n-1/2]
```

2: $\int \mathrm{Tan}[c+d\,x]^m\,(a+b\,\mathrm{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2\neq 0\ \land\ \frac{m}{2}\in\mathbb{Z}^-\ \land\ n-\frac{1}{2}\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: If $\frac{m}{2}\in\mathbb{Z}$, then $\mathrm{Tan}[z]^m == \left(-1+\mathrm{Csc}[z]^2\right)^{-m/2}$

Note: Note need find rules so restriction limiting m equal 2 can be lifted.

Rule: If $a^2-b^2\neq 0\ \land\ \frac{m}{2}\in\mathbb{Z}^-\ \land\ n-\frac{1}{2}\in\mathbb{Z}$, then

$$\int \mathrm{Tan}[c+d\,x]^m\,(a+b\,\mathrm{Sec}[c+d\,x])^n\,dx\ \longrightarrow\ \int (a+b\,\mathrm{Sec}[c+d\,x])^n\ \mathrm{ExpandIntegrand}\left[\left(-1+\mathrm{Csc}[c+d\,x]^2\right)^{-m/2},\,x\right]dx$$

Program code:

```
Int[cot[c_.+d_.*x_]^m_*(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(a+b*Csc[c+d*x])^n,(-1+Sec[c+d*x]^2)^(-m/2),x],x] /;
FreeQ[{a,b,c,d,n},x] && NeQ[a^2-b^2,0] && ILtQ[m/2,0] && IntegerQ[n-1/2] && EqQ[m,-2]
```

3: $\int (e\,\mathrm{Tan}[c+d\,x])^m\,(a+b\,\mathrm{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2\neq 0\,\land\,n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $a^2-b^2\neq 0\,\land\,n\in\mathbb{Z}^+$, then

$$\int (e\,\mathrm{Tan}[c+d\,x])^m\,(a+b\,\mathrm{Sec}[c+d\,x])^n\,dx\,\longrightarrow\,\int (e\,\mathrm{Tan}[c+d\,x])^m\,\mathrm{ExpandIntegrand}\left[(a+b\,\mathrm{Sec}[c+d\,x])^n,\,x\right]dx$$

Program code:

```
Int[(e_.*cot[c_.+d_.*x_])^m_*(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[ExpandIntegrand[(e*Cot[c+d*x])^m,(a+b*Csc[c+d*x])^n,x],x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[a^2-b^2,0] && IGtQ[n,0]
```

4: $\int \text{Tan}[c+d\,x]^m\,(a+b\,\text{Sec}[c+d\,x])^n\,dx$ when $a^2-b^2\neq 0\,\land\,n\in\mathbb{Z}\,\land\,m\in\mathbb{Z}\,\land\,\left(\frac{m}{2}\in\mathbb{Z}\,\lor\,m\leq 1\right)$

Derivation: Algebraic normalization

Basis: $a+b\,\text{Sec}[z]==\frac{b+a\,\text{Cos}[z]}{\text{Cos}[z]}$

Basis: $\text{Tan}[z]==\frac{\text{Sin}[z]}{\text{Cos}[z]}$

Rule: If $a^2-b^2\neq 0\,\land\,n\in\mathbb{Z}\,\land\,m\in\mathbb{Z}\,\land\,\left(\frac{m}{2}\in\mathbb{Z}\,\lor\,m\leq 1\right)$, then

$$\int \text{Tan}[c+d\,x]^m\,(a+b\,\text{Sec}[c+d\,x])^n\,dx\;\longrightarrow\;\int\frac{\text{Sin}[c+d\,x]^m\,(b+a\,\text{Cos}[c+d\,x])^n}{\text{Cos}[c+d\,x]^{m+n}}\,dx$$

Program code:

```
Int[cot[c_.+d_.*x_]^m_.*(a_+b_.*csc[c_.+d_.*x_])^n_,x_Symbol] :=
  Int[Cos[c+d*x]^m*(b+a*Sin[c+d*x])^n/Sin[c+d*x]^(m+n),x] /;
FreeQ[{a,b,c,d},x] && NeQ[a^2-b^2,0] && IntegerQ[n] && IntegerQ[m] && (IntegerQ[m/2] || LeQ[m,1])
```

U: $\int (e \operatorname{Tan}[c+d x])^m (a+b \operatorname{Sec}[c+d x])^n \, dx$

Rule:

$$\int (e \operatorname{Tan}[c+d x])^m (a+b \operatorname{Sec}[c+d x])^n \, dx \longrightarrow \int (e \operatorname{Tan}[c+d x])^m (a+b \operatorname{Sec}[c+d x])^n \, dx$$

Program code:

```
Int[(e_.*cot[c_.+d_.*x_])^m_.*(a_.+b_.*csc[c_.+d_.*x_])^n_.,x_Symbol] :=
  Unintegrable[(e*Cot[c+d*x])^m*(a+b*Csc[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,e,m,n},x]
```

Rules for integrands of the form $(d\,\mathrm{Tan}[e+f\,x]^p)^n\,(a+b\,\mathrm{Sec}[e+f\,x])^m$

1: $\int (e\,\mathrm{Tan}[c+d\,x]^p)^m\,(a+b\,\mathrm{Sec}[c+d\,x])^n\,dx$ when $m \notin \mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{(e\,\mathrm{Tan}[c+d\,x]^p)^m}{(e\,\mathrm{Tan}[c+d\,x])^{m\,p}} == 0$

Rule: If $m \notin \mathbb{Z}$, then

$$\int (e\,\mathrm{Tan}[c+d\,x]^p)^m\,(a+b\,\mathrm{Sec}[c+d\,x])^n\,dx \to \frac{(e\,\mathrm{Tan}[c+d\,x]^p)^m}{(e\,\mathrm{Tan}[c+d\,x])^{m\,p}}\int (e\,\mathrm{Tan}[c+d\,x])^{m\,p}\,(a+b\,\mathrm{Sec}[c+d\,x])^n\,dx$$

Program code:

```
Int[(e_.*cot[c_.+d_.*x_])^m_*(a_+b_.*sec[c_.+d_.*x_])^n_.,x_Symbol] :=
  (e*Cot[c+d*x])^m*Tan[c+d*x]^m*Int[(a+b*Sec[c+d*x])^n/Tan[c+d*x]^m,x] /;
FreeQ[{a,b,c,d,e,m,n},x] && Not[IntegerQ[m]]
```

```
Int[(e_.*tan[c_.+d_.*x_]^p_)^m_*(a_+b_.*sec[c_.+d_.*x_])^n_.,x_Symbol] :=
  (e*Tan[c+d*x]^p)^m/(e*Tan[c+d*x])^(m*p)*Int[(e*Tan[c+d*x])^(m*p)*(a+b*Sec[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && Not[IntegerQ[m]]
```

```
Int[(e_.*cot[c_.+d_.*x_]^p_)^m_*(a_+b_.*csc[c_.+d_.*x_])^n_.,x_Symbol] :=
  (e*Cot[c+d*x]^p)^m/(e*Cot[c+d*x])^(m*p)*Int[(e*Cot[c+d*x])^(m*p)*(a+b*Csc[c+d*x])^n,x] /;
FreeQ[{a,b,c,d,e,m,n,p},x] && Not[IntegerQ[m]]
```