# Rules for integrands of the form $(f\,x)^m\,(d+e\,x^{n-q})\,\left(a\,x^q+b\,x^n+c\,x^{2\,n-q}\right)^p$

**1:** $\int x^m\,\left(A+B\,x^{n-q}\right)\,\left(a\,x^q+b\,x^n+c\,x^{2\,n-q}\right)^p\,dx$ when $p\in\mathbb{Z}$

Rule: If $p\in\mathbb{Z}$, then

$$\int x^m\,\left(A+B\,x^{n-q}\right)\,\left(a\,x^q+b\,x^n+c\,x^{2\,n-q}\right)^p\,dx\;\longrightarrow\;\int x^{m+p\,q}\,\left(A+B\,x^{n-q}\right)\,\left(a+b\,x^{n-q}+c\,x^{2\,(n-q)}\right)^p\,dx$$

Program code:

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  Int[x^(m+p*q)*(A+B*x^(n-q))*(a+b*x^(n-q)+c*x^(2*(n-q)))^p,x] /;
FreeQ[{a,b,c,A,B,m,n,q},x] && EqQ[r,n-q] && EqQ[j,2*n-q] && IntegerQ[p] && PosQ[n-q]
```

**2.** $\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+$

**1:** $\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+ \land p > 0 \land m + p q \leq -(n-q) \land m + p q + 1 \neq 0 \land m + p q + (n-q)(2p+1) + 1 \neq 0$

Derivation: Generalized trinomial recurrence 1a

Rule: If $p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+ \land p > 0 \land$ $m + p q \leq -(n-q) \land m + p q + 1 \neq 0 \land m + p q + (n-q)(2p+1) + 1 \neq 0$, then

$$\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \longrightarrow$$

$$\left(\left(x^{m+1} \left(A (m + p q + (n-q)(2p+1) + 1) + B (m + p q + 1) x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p\right) / \left((m + p q + 1)(m + p q + (n-q)(2p+1) + 1)\right)\right) + \frac{(n-q)\, p}{(m + p q + 1)(m + p q + (n-q)(2p+1) + 1)} \cdot$$
$$\int x^{m+n} \left(2 a B (m + p q + 1) - A b (m + p q + (n-q)(2p+1) + 1) + \left(b B (m + p q + 1) - 2 A c (m + p q + (n-q)(2p+1) + 1)\right) x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^{p-1} dx$$

Program code:

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  x^(m+1)*(A*(m+p*q+(n-q)*(2*p+1)+1)+B*(m+p*q+1)*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^p/((m+p*q+1)*(m+p*q+(n-q)*(2*p+1)+1)) +
  (n-q)*p/((m+p*q+1)*(m+p*q+(n-q)*(2*p+1)+1))*
    Int[x^(n+m)*
      Simp[2*a*B*(m+p*q+1)-A*b*(m+p*q+(n-q)*(2*p+1)+1)+(b*B*(m+p*q+1)-2*A*c*(m+p*q+(n-q)*(2*p+1)+1))*x^(n-q),x]*
      (a*x^q+b*x^n+c*x^(2*n-q))^(p-1),x] /;
FreeQ[{a,b,c,A,B},x] && EqQ[r,n-q] && EqQ[j,2*n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[p,0] &&
  RationalQ[m,q] && LeQ[m+p*q,-(n-q)] && NeQ[m+p*q+1,0] && NeQ[m+p*q+(n-q)*(2*p+1)+1,0]
```

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  With[{n=q+r},
  x^(m+1)*(A*(m+p*q+(n-q)*(2*p+1)+1)+B*(m+p*q+1)*x^(n-q))*(a*x^q+c*x^(2*n-q))^p/((m+p*q+1)*(m+p*q+(n-q)*(2*p+1)+1)) +
  2*(n-q)*p/((m+p*q+1)*(m+p*q+(n-q)*(2*p+1)+1))*
    Int[x^(n+m)*Simp[a*B*(m+p*q+1)-A*c*(m+p*q+(n-q)*(2*p+1)+1)*x^(n-q),x]*(a*x^q+c*x^(2*n-q))^(p-1),x] /;
 EqQ[j,2*n-q] && IGtQ[n,0] && LeQ[m+p*q,-(n-q)] && NeQ[m+p*q+1,0] && NeQ[m+p*q+(n-q)*(2*p+1)+1,0]] /;
FreeQ[{a,c,A,B},x] && Not[IntegerQ[p]] && RationalQ[m,p,q] && GtQ[p,0]
```

**2:** $\int x^m \left(A+B x^{n-q}\right) \left(a x^q+b x^n+c x^{2 n-q}\right)^p dx$ when $p \notin \mathbb{Z} \land b^2-4 a c \neq 0 \land n \in \mathbb{Z}^+ \land p<-1 \land m+p q>n-q-1$

## Derivation: Generalized trinomial recurrence 2a

## Rule: If $p \notin \mathbb{Z} \land b^2-4 a c \neq 0 \land n \in \mathbb{Z}^+ \land p<-1 \land m+p q>n-q-1$, then

$$\int x^m \left(A+B x^{n-q}\right) \left(a x^q+b x^n+c x^{2 n-q}\right)^p dx \longrightarrow$$

$$\frac{x^{m-n+1} \left(A b-2 a B-\left(b B-2 A c\right) x^{n-q}\right) \left(a x^q+b x^n+c x^{2 n-q}\right)^{p+1}}{(n-q)\,(p+1)\,\left(b^2-4 a c\right)}+\frac{1}{(n-q)\,(p+1)\,\left(b^2-4 a c\right)}\cdot$$

$$\int x^{m-n} \left((m+p q-n+q+1)\,(2 a B-A b)+(m+p q+2\,(n-q)\,(p+1)+1)\,\left(b B-2 A c\right) x^{n-q}\right) \left(a x^q+b x^n+c x^{2 n-q}\right)^{p+1} dx$$

## Program code:

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  x^(m-n+1)*(A*b-2*a*B-(b*B-2*A*c)*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/((n-q)*(p+1)*(b^2-4*a*c)) +
  1/((n-q)*(p+1)*(b^2-4*a*c))*
    Int[x^(m-n)*
      Simp[(m+p*q-n+q+1)*(2*a*B-A*b)+(m+p*q+2*(n-q)*(p+1)+1)*(b*B-2*A*c)*x^(n-q),x]*
      (a*x^q+b*x^n+c*x^(2*n-q))^(p+1),x] /;
FreeQ[{a,b,c,A,B},x] && EqQ[r,n-q] && EqQ[j,2*n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1] &&
  RationalQ[m,q] && GtQ[m+p*q,n-q-1]
```

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  With[{n=q+r},
  x^(m-n+1)*(a*B-A*c*x^(n-q))*(a*x^q+c*x^(2*n-q))^(p+1)/(2*a*c*(n-q)*(p+1)) -
  1/(2*a*c*(n-q)*(p+1))*
     Int[x^(m-n)*Simp[a*B*(m+p*q-n+q+1)-A*c*(m+p*q+(n-q)*2*(p+1)+1)*x^(n-q),x]*(a*x^q+c*x^(2*n-q))^(p+1),x] /;
 EqQ[j,2*n-q] && IGtQ[n,0] && m+p*q>n-q-1] /;
FreeQ[{a,c,A,B},x] && Not[IntegerQ[p]] && RationalQ[m,q] && LtQ[p,-1]
```

**3:** $\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p \, dx$ when
$p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+ \land p > 0 \land m + p q > -(n-q) - 1 \land m + p (2n-q) + 1 \neq 0 \land m + p q + (n-q)(2p+1) + 1 \neq 0$

## Derivation: Generalized trinomial recurrence 1b

## Rule: If $p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+ \land p > 0 \land m + p q > -(n-q) - 1 \land$, then $m + p (2n-q) + 1 \neq 0 \land m + p q + (n-q)(2p+1) + 1 \neq 0$

$$\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \longrightarrow$$

$$\left(\left(x^{m+1} \left(b B (n-q) p + A c (m + p q + (n-q)(2p+1) + 1) + B c (m + p (2n-q) + 1) x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p\right) \middle/ \left(c (m + p (2n-q) + 1)(m + p q + (n-q)(2p+1) + 1)\right)\right) +$$
$$\frac{(n-q) p}{c (m + p (2n-q) + 1)(m + p q + (n-q)(2p+1) + 1)} \int x^{m+q} \left(2 a A c (m + p q + (n-q)(2p+1) + 1) - a b B (m + p q + 1) + \right.$$
$$\left.\left(2 a B c (m + p (2n-q) + 1) + A b c (m + p q + (n-q)(2p+1) + 1) - b^2 B (m + p q + (n-q) p + 1)\right) x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^{p-1} dx$$

## Program code:

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  x^(m+1)*(b*B*(n-q)*p+A*c*(m+p*q+(n-q)*(2*p+1)+1)+B*c*(m+p*q+2*(n-q)*p+1)*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^p/
    (c*(m+p*(2*n-q)+1)*(m+p*q+(n-q)*(2*p+1)+1)) +
  (n-q)*p/(c*(m+p*(2*n-q)+1)*(m+p*q+(n-q)*(2*p+1)+1))*
    Int[x^(m+q)*
      Simp[2*a*A*c*(m+p*q+(n-q)*(2*p+1)+1)-a*b*B*(m+p*q+1)+
        (2*a*B*c*(m+p*q+2*(n-q)*p+1)+A*b*c*(m+p*q+(n-q)*(2*p+1)+1)-b^2*B*(m+p*q+(n-q)*p+1))*x^(n-q),x]*
      (a*x^q+b*x^n+c*x^(2*n-q))^(p-1),x] /;
FreeQ[{a,b,c,A,B},x] && EqQ[r,n-q] && EqQ[j,2*n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GtQ[p,0] &&
  RationalQ[m,q] && GtQ[m+p*q,-(n-q)-1] && NeQ[m+p*(2*n-q)+1,0] && NeQ[m+p*q+(n-q)*(2*p+1)+1,0]
```

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  With[{n=q+r},
  x^(m+1)*(A*(m+p*q+(n-q)*(2*p+1)+1)+B*(m+p*q+2*(n-q)*p+1)*x^(n-q))*(a*x^q+c*x^(2*n-q))^p/
    ((m+p*(2*n-q)+1)*(m+p*q+(n-q)*(2*p+1)+1)) +
   (n-q)*p/((m+p*(2*n-q)+1)*(m+p*q+(n-q)*(2*p+1)+1))*
     Int[x^(m+q)*Simp[2*a*A*(m+p*q+(n-q)*(2*p+1)+1)+2*a*B*(m+p*q+2*(n-q)*p+1)*x^(n-q),x]*(a*x^q+c*x^(2*n-q))^(p-1),x] /;
 EqQ[j,2*n-q] && IGtQ[n,0] && GtQ[m+p*q,-(n-q)] && NeQ[m+p*q+2*(n-q)*p+1,0] && NeQ[m+p*q+(n-q)*(2*p+1)+1,0] && NeQ[m+1,n]] /;
FreeQ[{a,c,A,B},x] && Not[IntegerQ[p]] && RationalQ[m,q] && GtQ[p,0]
```

**4:** $\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+ \land p < -1 \land m + p q < n - q - 1$

## Derivation: Generalized trinomial recurrence 2b

## Rule: If $p \notin \mathbb{Z} \land b^2 - 4 a c \neq 0 \land n \in \mathbb{Z}^+ \land p < -1 \land m + p q < n - q - 1$, then

$$\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \longrightarrow$$

$$-\left(\left(x^{m-q+1} \left(A b^2 - a b B - 2 a A c + \left(A b - 2 a B\right) c x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1}\right) \Big/ \left(a (n-q) (p+1) \left(b^2 - 4 a c\right)\right)\right) +$$
$$\frac{1}{a (n-q) (p+1) \left(b^2 - 4 a c\right)} \int x^{m-q} \left(A b^2 (m + p q + (n-q) (p+1) + 1) - a b B (m + p q + 1) - 2 a A c (m + p q + 2 (n-q) (p+1) + 1) + (m + p q + (n-q) (2 p + 3) + 1) \left(A b - 2 a B\right) c x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1} dx$$

## Program code:

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  -x^(m-q+1)*(A*b^2-a*b*B-2*a*A*c+(A*b-2*a*B)*c*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/(a*(n-q)*(p+1)*(b^2-4*a*c)) +
  1/(a*(n-q)*(p+1)*(b^2-4*a*c))*
    Int[x^(m-q)*
      Simp[A*b^2*(m+p*q+(n-q)*(p+1)+1)-a*b*B*(m+p*q+1)-2*a*A*c*(m+p*q+2*(n-q)*(p+1)+1)+
        (m+p*q+(n-q)*(2*p+3)+1)*(A*b-2*a*B)*c*x^(n-q),x]*
      (a*x^q+b*x^n+c*x^(2*n-q))^(p+1),x] /;
FreeQ[{a,b,c,A,B},x] && EqQ[r,n-q] && EqQ[j,2*n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && LtQ[p,-1] &&
  RationalQ[m,q] && m+p*q<n-q-1
```

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  With[{n=q+r},
  -x^(m-q+1)*(A*c+B*c*x^(n-q))*(a*x^q+c*x^(2*n-q))^(p+1)/(2*a*c*(n-q)*(p+1)) +
  1/(2*a*c*(n-q)*(p+1))*
    Int[x^(m-q)*Simp[A*c*(m+p*q+2*(n-q)*(p+1)+1)+B*(m+p*q+(n-q)*(2*p+3)+1)*c*x^(n-q),x]*(a*x^q+c*x^(2*n-q))^(p+1),x] /;
 EqQ[j,2*n-q] && IGtQ[n,0] && LtQ[m+p*q,n-q-1]] /;
FreeQ[{a,c,A,B},x] && Not[IntegerQ[p]] && RationalQ[m,q] && LtQ[p,-1]
```

5: $\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$ when $p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land -1 \leq p < 0 \land m + pq \geq n - q - 1 \land m + pq + (n-q)(2p+1) + 1 \neq 0$

## Derivation: Generalized trinomial recurrence 3a

## Rule: If

$p \notin \mathbb{Z} \land b^2 - 4ac \neq 0 \land n \in \mathbb{Z}^+ \land -1 \leq p < 0 \land m + pq \geq n - q - 1 \land m + pq + (n-q)(2p+1) + 1 \neq 0$, then

$$\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p dx \longrightarrow$$

$$\frac{B x^{m-n+1} \left(a x^q + b x^n + c x^{2n-q}\right)^{p+1}}{c (m + pq + (n-q)(2p+1) + 1)} - \frac{1}{c (m + pq + (n-q)(2p+1) + 1)} \cdot$$

$$\int x^{m-n+q} \left(a B (m + pq - n + q + 1) + \left(b B (m + pq + (n-q) p + 1) - A c (m + pq + (n-q)(2p+1) + 1)\right) x^{n-q}\right) \left(a x^q + b x^n + c x^{2n-q}\right)^p dx$$

## Program code:

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  B*x^(m-n+1)*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/(c*(m+p*q+(n-q)*(2*p+1)+1)) -
  1/(c*(m+p*q+(n-q)*(2*p+1)+1))*
    Int[x^(m-n+q)*
      Simp[a*B*(m+p*q-n+q+1)+(b*B*(m+p*q+(n-q)*p+1)-A*c*(m+p*q+(n-q)*(2*p+1)+1))*x^(n-q),x]*
      (a*x^q+b*x^n+c*x^(2*n-q))^p,x] /;
FreeQ[{a,b,c,A,B},x] && EqQ[r,n-q] && EqQ[j,2*n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] && GeQ[p,-1] && LtQ[p,0] &&
  RationalQ[m,q] && GeQ[m+p*q,n-q-1] && NeQ[m+p*q+(n-q)*(2*p+1)+1,0]
```

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  With[{n=q+r},
  B*x^(m-n+1)*(a*x^q+c*x^(2*n-q))^(p+1)/(c*(m+p*q+(n-q)*(2*p+1)+1)) -
  1/(c*(m+p*q+(n-q)*(2*p+1)+1))*
    Int[x^(m-n+q)*Simp[a*B*(m+p*q-n+q+1)-A*c*(m+p*q+(n-q)*(2*p+1)+1)*x^(n-q),x]*(a*x^q+c*x^(2*n-q))^p,x] /;
 EqQ[j,2*n-q] && IGtQ[n,0] && GeQ[m+p*q,n-q-1] && NeQ[m+p*q+(n-q)*(2*p+1)+1,0]] /;
FreeQ[{a,c,A,B},x] && Not[IntegerQ[p]] && RationalQ[m,p,q] && GeQ[p,-1] && LtQ[p,0]
```

**6:** $\int x^m \left(A+B x^{n-q}\right) \left(a x^q+b x^n+c x^{2n-q}\right)^p \, dx$ when $p \notin \mathbb{Z} \land b^2-4ac \neq 0 \land n \in \mathbb{Z}^+ \land -1 \leq p < 0 \land m+pq \leq -(n-q) \land m+pq+1 \neq 0$

## Derivation: Generalized trinomial recurrence 3b

## Rule: If $p \notin \mathbb{Z} \land b^2-4ac \neq 0 \land n \in \mathbb{Z}^+ \land m+pq \leq -(n-q) \land -1 \leq p < 0 \land m+pq+1 \neq 0$, then

$$\int x^m \left(A+B x^{n-q}\right) \left(a x^q+b x^n+c x^{2n-q}\right)^p \, dx \to$$

$$\frac{A x^{m-q+1} \left(a x^q+b x^n+c x^{2n-q}\right)^{p+1}}{a (m+pq+1)}+\frac{1}{a (m+pq+1)}\cdot$$

$$\int x^{m+n-q} \left(a B (m+pq+1)-A b (m+pq+(n-q)(p+1)+1)-A c (m+pq+2(n-q)(p+1)+1) x^{n-q}\right) \left(a x^q+b x^n+c x^{2n-q}\right)^p \, dx$$

## Program code:

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  A*x^(m-q+1)*(a*x^q+b*x^n+c*x^(2*n-q))^(p+1)/(a*(m+p*q+1)) +
  1/(a*(m+p*q+1))*
    Int[x^(m+n-q)*
      Simp[a*B*(m+p*q+1)-A*b*(m+p*q+(n-q)*(p+1)+1)-A*c*(m+p*q+2*(n-q)*(p+1)+1)*x^(n-q),x]*
      (a*x^q+b*x^n+c*x^(2*n-q))^p,x] /;
FreeQ[{a,b,c,A,B},x] && EqQ[r,n-q] && EqQ[j,2*n-q] && Not[IntegerQ[p]] && NeQ[b^2-4*a*c,0] && IGtQ[n,0] &&
  RationalQ[m,p,q] && (GeQ[p,-1] && LtQ[p,0] || EqQ[m+p*q+(n-q)*(2*p+1)+1,0]) && LeQ[m+p*q,-(n-q)] && NeQ[m+p*q+1,0]
```

```
Int[x_^m_.*(A_+B_.*x_^r_.)*(a_.*x_^q_.+c_.*x_^j_.)^p_.,x_Symbol] :=
  With[{n=q+r},
  A*x^(m-q+1)*(a*x^q+c*x^(2*n-q))^(p+1)/(a*(m+p*q+1)) +
  1/(a*(m+p*q+1))*
    Int[x^(m+n-q)*Simp[a*B*(m+p*q+1)-A*c*(m+p*q+2*(n-q)*(p+1)+1)*x^(n-q),x]*(a*x^q+c*x^(2*n-q))^p,x] /;
 EqQ[j,2*n-q] && IGtQ[n,0] && (GeQ[p,-1] && LtQ[p,0] || EqQ[m+p*q+(n-q)*(2*p+1)+1,0]) && LeQ[m+p*q,-(n-q)] && NeQ[m+p*q+1,0]] /;
FreeQ[{a,c,A,B},x] && Not[IntegerQ[p]] && RationalQ[m,p,q]
```

**3:** $\int \frac{x^m \left(A + B x^{n-q}\right)}{\sqrt{a x^q + b x^n + c x^{2 n-q}}} \, dx$ when $q < n$

## Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{q/2} \sqrt{a+b x^{n-q}+c x^{2 (n-q)}}}{\sqrt{a x^q+b x^n+c x^{2 n-q}}} == 0$

Rule: If $q < n$, then

$$\int \frac{x^m \left(A + B x^{n-q}\right)}{\sqrt{a x^q + b x^n + c x^{2 n-q}}} \, dx \longrightarrow \frac{x^{q/2} \sqrt{a + b x^{n-q} + c x^{2 (n-q)}}}{\sqrt{a x^q + b x^n + c x^{2 n-q}}} \int \frac{x^{m-q/2} \left(A + B x^{n-q}\right)}{\sqrt{a + b x^{n-q} + c x^{2 (n-q)}}} \, dx$$

## Program code:

```
Int[x_^m_.*(A_+B_.*x_^j_.)/Sqrt[a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.],x_Symbol] :=
  x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))]/Sqrt[a*x^q+b*x^n+c*x^(2*n-q)]*
    Int[x^(m-q/2)*(A+B*x^(n-q))/Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))],x] /;
FreeQ[{a,b,c,A,B,m,n,q},x] && EqQ[j,n-q] && EqQ[r,2*n-q] && PosQ[n-q] &&
    (EqQ[m,1/2] || EqQ[m,-1/2]) && EqQ[n,3] && EqQ[q,1]
```

**x.** $\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2 n-q}\right)^p dx$ when $p + \frac{1}{2} \in \mathbb{Z}$

**x:** $\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2 n-q}\right)^p dx$ when $p + \frac{1}{2} \in \mathbb{Z}^+$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\sqrt{a x^q+b x^n+c x^{2 n-q}}}{x^{q/2} \sqrt{a+b x^{n-q}+c x^{2 (n-q)}}} == 0$

Rule: If $p + \frac{1}{2} \in \mathbb{Z}^+$, then

$$\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2 n-q}\right)^p dx \rightarrow \frac{\sqrt{a x^q + b x^n + c x^{2 n-q}}}{x^{q/2} \sqrt{a + b x^{n-q} + c x^{2 (n-q)}}} \int x^{m+q p} \left(A + B x^{n-q}\right) \left(a + b x^{n-q} + c x^{2 (n-q)}\right)^p dx$$

Program code:

```
(* Int[x_^m_.*(A_+B_.*x_^j_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  Sqrt[a*x^q+b*x^n+c*x^(2*n-q)]/(x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))])*
    Int[x^(m+q*p)*(A+B*x^(n-q))*(a+b*x^(n-q)+c*x^(2*(n-q)))^p,x] /;
FreeQ[{a,b,c,A,B,m,n,p,q},x] && EqQ[j,n-q] && EqQ[r,2*n-q] && IGtQ[p+1/2,0] && PosQ[n-q] *)
```

**x:** $\int x^m \left(A + B x^{n-q}\right) \left(a x^q + b x^n + c x^{2 n-q}\right)^p dx$ when $p - \frac{1}{2} \in \mathbb{Z}^-$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{x^{q/2} \sqrt{a+b x^{n-q}+c x^{2 (n-q)}}}{\sqrt{a x^q+b x^n+c x^{2 n-q}}} == 0$

Rule: If $p - \frac{1}{2} \in \mathbb{Z}^-$, then

$$\int x^m \left(A+B x^{n-q}\right) \left(a x^q+b x^n+c x^{2 n-q}\right)^p dx \longrightarrow \frac{x^{q/2} \sqrt{a+b x^{n-q}+c x^{2 (n-q)}}}{\sqrt{a x^q+b x^n+c x^{2 n-q}}} \int x^{m+q p} \left(A+B x^{n-q}\right) \left(a+b x^{n-q}+c x^{2 (n-q)}\right)^p dx$$

## Program code:

```
(* Int[x_^m_.*(A_+B_.*x_^j_.)*(a_.*x_^q_.+b_.*x_^n_.+c_.*x_^r_.)^p_,x_Symbol] :=
  x^(q/2)*Sqrt[a+b*x^(n-q)+c*x^(2*(n-q))]/Sqrt[a*x^q+b*x^n+c*x^(2*n-q)]*
    Int[x^(m+q*p)*(A+B*x^(n-q))*(a+b*x^(n-q)+c*x^(2*(n-q)))^p,x] /;
FreeQ[{a,b,c,A,B,m,n,p,q},x] && EqQ[j,n-q] && EqQ[r,2*n-q] && ILtQ[p-1/2,0] && PosQ[n-q] *)
```

**4:** $\int x^m \left(A+B x^{k-j}\right) \left(a x^j+b x^k+c x^{2 k-j}\right)^p dx$ when $p \notin \mathbb{Z}$

## Derivation: Piecewise constant extraction

## Basis: $\partial_x \frac{\left(a x^j+b x^k+c x^{2 k-j}\right)^p}{x^{j p} \left(a+b x^{k-j}+c x^{2 (k-j)}\right)^p} == 0$

## Rule: If $p \notin \mathbb{Z}$, then

$$\int x^m \left(A+B x^{k-j}\right) \left(a x^j+b x^k+c x^{2 k-j}\right)^p dx \longrightarrow \frac{\left(a x^j+b x^k+c x^{2 k-j}\right)^p}{x^{j p} \left(a+b x^{k-j}+c x^{2 (k-j)}\right)^p} \int x^{m+j p} \left(A+B x^{k-j}\right) \left(a+b x^{k-j}+c x^{2 (k-j)}\right)^p dx$$

## Program code:

```
Int[x_^m_.*(A_+B_.*x_^q_)*(a_.*x_^j_.+b_.*x_^k_.+c_.*x_^n_.)^p_,x_Symbol] :=
  (a*x^j+b*x^k+c*x^n)^p/(x^(j*p)*(a+b*x^(k-j)+c*x^(2*(k-j)))^p)*
    Int[x^(m+j*p)*(A+B*x^(k-j))*(a+b*x^(k-j)+c*x^(2*(k-j)))^p,x] /;
FreeQ[{a,b,c,A,B,j,k,m,p},x] && EqQ[q,k-j] && EqQ[n,2*k-j] && Not[IntegerQ[p]] && PosQ[k-j]
```

**S:** $\int u^m \left(A + B\,u^{n-q}\right) \left(a\,u^q + b\,u^n + c\,u^{2\,n-q}\right)^p\,dx$ when $u == d + e\,x$

## Derivation: Integration by substitution

## Rule: If $u == d + e\,x$, then

$$\int u^m \left(A + B\,u^{n-q}\right) \left(a\,u^q + b\,u^n + c\,u^{2\,n-q}\right)^p\,dx \rightarrow \frac{1}{e}\,\text{Subst}\left[\int x^m \left(A + B\,x^{n-q}\right) \left(a\,x^q + b\,x^n + c\,x^{2\,n-q}\right)^p\,dx,\ x,\ u\right]$$

## Program code:

```
Int[u_^m_.*(A_+B_.*u_^j_.)*(a_.*u_^q_.+b_.*u_^n_.+c_.*u_^r_.)^p_.,x_Symbol] :=
  1/Coefficient[u,x,1]*Subst[Int[x^m*(A+B*x^(n-q))*(a*x^q+b*x^n+c*x^(2*n-q))^p,x],x,u] /;
FreeQ[{a,b,c,A,B,m,n,p,q},x] && EqQ[j,n-q] && EqQ[r,2*n-q] && LinearQ[u,x] && NeQ[u,x]
```