

## Rules for integrands of the form $(dx)^m (a + b \operatorname{ArcSin}[cx])^n$

1.  $\int (dx)^m (a + b \operatorname{ArcSin}[cx])^n dx$  when  $n \in \mathbb{Z}^+$

1:  $\int \frac{(a + b \operatorname{ArcSin}[cx])^n}{x} dx$  when  $n \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis:  $\frac{F[\operatorname{ArcSin}[cx]]}{x} = \operatorname{Subst}\left[\frac{F[x]}{\operatorname{Tan}[x]}, x, \operatorname{ArcSin}[cx]\right] \partial_x \operatorname{ArcSin}[cx]$

Note:  $\frac{(a+bx)^n}{\operatorname{Tan}[x]}$  is not integrable unless  $n \in \mathbb{Z}^+$ .

Rule: If  $n \in \mathbb{Z}^+$ , then

$$\int \frac{(a + b \operatorname{ArcSin}[cx])^n}{x} dx \rightarrow \operatorname{Subst}\left[\int \frac{(a + bx)^n}{\operatorname{Tan}[x]} dx, x, \operatorname{ArcSin}[cx]\right]$$

Program code:

```
Int[(a_.+b_.*ArcSin[c_.*x_])^n_/x_,x_Symbol] :=
  Subst[Int[(a+b*x)^n/Tan[x],x],x,ArcSin[c*x] /;
  FreeQ[{a,b,c},x] && IGtQ[n,0]
```

```
Int[(a_.+b_.*ArcCos[c_.*x_])^n_/x_,x_Symbol] :=
  -Subst[Int[(a+b*x)^n/Cot[x],x],x,ArcCos[c*x] /;
  FreeQ[{a,b,c},x] && IGtQ[n,0]
```

2:  $\int (dx)^m (a + b \operatorname{ArcSin}[cx])^n dx$  when  $n \in \mathbb{Z}^+ \wedge m \neq -1$

Reference: G&R 2.831, CRC 453, A&S 4.4.65

Reference: G&R 2.832, CRC 454, A&S 4.4.67

Derivation: Integration by parts

$$\text{Basis: } \partial_x (a + b \operatorname{ArcSin}[cx])^n == \frac{bcn (a+b \operatorname{ArcSin}[cx])^{n-1}}{\sqrt{1-c^2 x^2}}$$

Rule: If  $n \in \mathbb{Z}^+ \wedge m \neq -1$ , then

$$\int (dx)^m (a + b \operatorname{ArcSin}[cx])^n dx \rightarrow \frac{(dx)^{m+1} (a + b \operatorname{ArcSin}[cx])^n}{d(m+1)} - \frac{bcn}{d(m+1)} \int \frac{(dx)^{m+1} (a + b \operatorname{ArcSin}[cx])^{n-1}}{\sqrt{1-c^2 x^2}} dx$$

Program code:

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcSin[c*x])^n/(d*(m+1)) -
  b*c*n/(d*(m+1))*Int[(d*x)^(m+1)*(a+b*ArcSin[c*x])^(n-1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && NeQ[m,-1]
```

```
Int[(d_.*x_)^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  (d*x)^(m+1)*(a+b*ArcCos[c*x])^n/(d*(m+1)) +
  b*c*n/(d*(m+1))*Int[(d*x)^(m+1)*(a+b*ArcCos[c*x])^(n-1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && NeQ[m,-1]
```

2.  $\int x^m (a + b \operatorname{ArcSin}[cx])^n dx$  when  $m \in \mathbb{Z}^+$

1:  $\int x^m (a + b \operatorname{ArcSin}[cx])^n dx$  when  $m \in \mathbb{Z}^+ \wedge n > 0$

Reference: G&R 2.831, CRC 453, A&S 4.4.65

Reference: G&R 2.832, CRC 454, A&S 4.4.67

Derivation: Integration by parts

$$\text{Basis: } \partial_x (a + b \operatorname{ArcSin}[cx])^n == \frac{bcn (a+b \operatorname{ArcSin}[cx])^{n-1}}{\sqrt{1-c^2 x^2}}$$

Rule: If  $n \in \mathbb{Z}^+ \wedge m \neq -1$ , then

$$\int x^m (a + b \operatorname{ArcSin}[c x])^n dx \rightarrow \frac{x^{m+1} (a + b \operatorname{ArcSin}[c x])^n}{m+1} - \frac{b c n}{m+1} \int \frac{x^{m+1} (a + b \operatorname{ArcSin}[c x])^{n-1}}{\sqrt{1 - c^2 x^2}} dx$$

## Program code:

```
Int[x_^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  x^(m+1)*(a+b*ArcSin[c*x])^n/(m+1) -
  b*c*n/(m+1)*Int[x^(m+1)*(a+b*ArcSin[c*x])^(n-1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GtQ[n,0]
```

```
Int[x_^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  x^(m+1)*(a+b*ArcCos[c*x])^n/(m+1) +
  b*c*n/(m+1)*Int[x^(m+1)*(a+b*ArcCos[c*x])^(n-1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GtQ[n,0]
```

$$2. \int x^m (a + b \operatorname{ArcSin}[cx])^n dx \text{ when } m \in \mathbb{Z}^+ \wedge n < -1$$

$$1: \int x^m (a + b \operatorname{ArcSin}[cx])^n dx \text{ when } m \in \mathbb{Z}^+ \wedge -2 \leq n < -1$$

Derivation: Integration by parts and integration by substitution

$$\text{Basis: } \frac{(a+b \operatorname{ArcSin}[cx])^n}{\sqrt{1-c^2 x^2}} == \partial_x \frac{(a+b \operatorname{ArcSin}[cx])^{n+1}}{b c (n+1)}$$

$$\text{Basis: } \frac{F[x]}{\sqrt{1-c^2 x^2}} == \frac{1}{c} \operatorname{Subst}[F[\frac{\operatorname{Sin}[x]}{c}], x, \operatorname{ArcSin}[cx]] \partial_x \operatorname{ArcSin}[cx]$$

Basis: If  $c > 0 \vee m \in \mathbb{Z}$ , then

$$\frac{x^{m-1} (m - (m+1) c^2 x^2)}{\sqrt{1-c^2 x^2}} == \frac{1}{c^m} \operatorname{Subst}[\operatorname{Sin}[x]^{m-1} (m - (m+1) \operatorname{Sin}[x]^2), x, \operatorname{ArcSin}[cx]] \partial_x \operatorname{ArcSin}[cx]$$

Note: Although not essential, by switching to the trig world this rule saves numerous steps and results in more compact antiderivatives.

Rule: If  $m \in \mathbb{Z}^+ \wedge -2 \leq n < -1$ , then

$$\begin{aligned} & \int x^m (a + b \operatorname{ArcSin}[cx])^n dx \rightarrow \\ & \frac{x^m \sqrt{1-c^2 x^2} (a + b \operatorname{ArcSin}[cx])^{n+1}}{b c (n+1)} - \frac{1}{b c (n+1)} \int \frac{x^{m-1} (m - (m+1) c^2 x^2) (a + b \operatorname{ArcSin}[cx])^{n+1}}{\sqrt{1-c^2 x^2}} dx \rightarrow \\ & \frac{x^m \sqrt{1-c^2 x^2} (a + b \operatorname{ArcSin}[cx])^{n+1}}{b c (n+1)} - \frac{1}{b c^{m+1} (n+1)} \operatorname{Subst}\left[\int (a + b x)^{n+1} \operatorname{Sin}[x]^{m-1} (m - (m+1) \operatorname{Sin}[x]^2) dx, x, \operatorname{ArcSin}[cx]\right] \end{aligned}$$

Program code:

```

Int[x_^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  x^m*Sqrt[1-c^2*x^2]*(a+b*ArcSin[c*x])^(n+1)/(b*c*(n+1)) -
  1/(b*c^(m+1)*(n+1))*Subst[Int[ExpandTrigReduce[(a+b*x)^(n+1),Sin[x]^(m-1)*(m-(m+1)*Sin[x]^2),x],x],x,ArcSin[c*x]] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GeQ[n,-2] && LtQ[n,-1]

```

```

Int[x_^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  -x^m*Sqrt[1-c^2*x^2]*(a+b*ArcCos[c*x])^(n+1)/(b*c*(n+1)) -
  1/(b*c^(m+1)*(n+1))*Subst[Int[ExpandTrigReduce[(a+b*x)^(n+1),Cos[x]^(m-1)*(m-(m+1)*Cos[x]^2),x],x],x,ArcCos[c*x]] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && GeQ[n,-2] && LtQ[n,-1]

```

$$\mathbf{2:} \int x^m (a + b \operatorname{ArcSin}[c x])^n dx \text{ when } m \in \mathbb{Z}^+ \wedge n < -2$$

Derivation: Integration by parts and algebraic expansion

$$\text{Basis: } \frac{(a+b \operatorname{ArcSin}[c x])^n}{\sqrt{1-c^2 x^2}} == \partial_x \frac{(a+b \operatorname{ArcSin}[c x])^{n+1}}{b c (n+1)}$$

$$\text{Basis: } \partial_x \left( x^m \sqrt{1-c^2 x^2} \right) == \frac{m x^{m-1}}{\sqrt{1-c^2 x^2}} - \frac{c^2 (m+1) x^{m+1}}{\sqrt{1-c^2 x^2}}$$

Rule: If  $m \in \mathbb{Z}^+ \wedge n < -2$ , then

$$\int x^m (a + b \operatorname{ArcSin}[c x])^n dx \rightarrow \frac{x^m \sqrt{1-c^2 x^2} (a + b \operatorname{ArcSin}[c x])^{n+1}}{b c (n+1)} - \frac{m}{b c (n+1)} \int \frac{x^{m-1} (a + b \operatorname{ArcSin}[c x])^{n+1}}{\sqrt{1-c^2 x^2}} dx + \frac{c (m+1)}{b (n+1)} \int \frac{x^{m+1} (a + b \operatorname{ArcSin}[c x])^{n+1}}{\sqrt{1-c^2 x^2}} dx$$

Program code:

```
Int[x_^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  x^m*Sqrt[1-c^2*x^2]*(a+b*ArcSin[c*x])^(n+1)/(b*c*(n+1)) -
  m/(b*c*(n+1))*Int[x^(m-1)*(a+b*ArcSin[c*x])^(n+1)/Sqrt[1-c^2*x^2],x] +
  c*(m+1)/(b*(n+1))*Int[x^(m+1)*(a+b*ArcSin[c*x])^(n+1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && LtQ[n,-2]
```

```
Int[x_^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  -x^m*Sqrt[1-c^2*x^2]*(a+b*ArcCos[c*x])^(n+1)/(b*c*(n+1)) +
  m/(b*c*(n+1))*Int[x^(m-1)*(a+b*ArcCos[c*x])^(n+1)/Sqrt[1-c^2*x^2],x] -
  c*(m+1)/(b*(n+1))*Int[x^(m+1)*(a+b*ArcCos[c*x])^(n+1)/Sqrt[1-c^2*x^2],x] /;
FreeQ[{a,b,c},x] && IGtQ[m,0] && LtQ[n,-2]
```

**3:**  $\int x^m (a + b \operatorname{ArcSin}[c x])^n dx$  when  $m \in \mathbb{Z}^+$

Derivation: Integration by substitution

Basis: If  $m \in \mathbb{Z}^+$ , then  $x^m F[\operatorname{ArcSin}[c x]] = \frac{1}{c^{m+1}} \operatorname{Subst}[F[x] \sin[x]^m \cos[x], x, \operatorname{ArcSin}[c x]] \partial_x \operatorname{ArcSin}[c x]$

Note: If  $m \in \mathbb{Z}^+$ , then  $(a + b x)^n \sin[x]^m \cos[x]$  is integrable in closed-form.

Rule: If  $m \in \mathbb{Z}^+$ , then

$$\int x^m (a + b \operatorname{ArcSin}[c x])^n dx \rightarrow \frac{1}{c^{m+1}} \operatorname{Subst}\left[\int (a + b x)^n \sin[x]^m \cos[x] dx, x, \operatorname{ArcSin}[c x]\right]$$

Program code:

```
Int[x^m_.*(a_.+b_.*ArcSin[c_.*x_])^n_,x_Symbol] :=
  1/c^(m+1)*Subst[Int[(a+b*x)^n*Sin[x]^m*Cos[x],x],x,ArcSin[c*x]] /;
FreeQ[{a,b,c,n},x] && IGtQ[m,0]
```

```
Int[x^m_.*(a_.+b_.*ArcCos[c_.*x_])^n_,x_Symbol] :=
  -1/c^(m+1)*Subst[Int[(a+b*x)^n*Cos[x]^m*Sin[x],x],x,ArcCos[c*x]] /;
FreeQ[{a,b,c,n},x] && IGtQ[m,0]
```

**U:**  $\int (d x)^m (a + b \operatorname{ArcSin}[c x])^n dx$

Rule:

$$\int (d x)^m (a + b \operatorname{ArcSin}[c x])^n dx \rightarrow \int (d x)^m (a + b \operatorname{ArcSin}[c x])^n dx$$

Program code:

```
Int[(d_.**x_)^m_.*(a_.+b_.*ArcSin[c_.**x_])^n_,x_Symbol] :=
  Unintegrable[(d*x)^m*(a+b*ArcSin[c*x])^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```

```
Int[(d_.**x_)^m_.*(a_.+b_.*ArcCos[c_.**x_])^n_,x_Symbol] :=
  Unintegrable[(d*x)^m*(a+b*ArcCos[c*x])^n,x] /;
FreeQ[{a,b,c,d,m,n},x]
```