

## Rules for integrands involving partial derivatives

1.  $\int u f^{(n)}[x] dx$

1:  $\int f^{(n)}[x] dx$

Reference: G&R 2.02.4

Rule:

$$\int f^{(n)}[x] dx \rightarrow f^{(n-1)}[x]$$

Program code:

```
Int[Derivative[n_][f_][x_],x_Symbol] :=
  Derivative[n-1][f][x] /;
FreeQ[{f,n},x]
```

2.  $\int (c F^{a+bx})^p f^{(n)}[x] dx$

1:  $\int (c F^{a+bx})^p f^{(n)}[x] dx$  when  $n \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If  $n \in \mathbb{Z}^+$ , then

$$\int (c F^{a+bx})^p f^{(n)}[x] dx \rightarrow (c F^{a+bx})^p f^{(n-1)}[x] - b p \log[F] \int (c F^{a+bx})^p f^{(n-1)}[x] dx$$

Program code:

```
Int[(c_.*F^(a_.+b_.*x_))^p_.*Derivative[n_][f_][x_],x_Symbol] :=
  (c*F^(a+b*x))^p*Derivative[n-1][f][x] - b*p*Log[F]*Int[(c*F^(a+b*x))^p*Derivative[n-1][f][x],x] /;
FreeQ[{a,b,c,f,F,p},x] && IGtQ[n,0]
```

**2:**  $\int (c F^{a+bx})^p f^{(n)}[x] dx$  when  $n \in \mathbb{Z}^-$

Derivation: Integration by parts

Rule: If  $n \in \mathbb{Z}^-$ , then

$$\int (c F^{a+bx})^p f^{(n)}[x] dx \rightarrow \frac{(c F^{a+bx})^p f^{(n)}[x]}{b p \log[F]} - \frac{1}{b p \log[F]} \int (c F^{a+bx})^p f^{(n+1)}[x] dx$$

Program code:

```
Int[(c_.*F^(a_.+b_.*x_))^p_.*Derivative[n_][f_][x_],x_Symbol] :=
  (c*F^(a+b*x))^p*Derivative[n][f][x]/(b*p*Log[F]) - 1/(b*p*Log[F])*Int[(c*F^(a+b*x))^p*Derivative[n+1][f][x],x] /;
FreeQ[{a,b,c,f,F,p},x] && ILtQ[n,0]
```

3.  $\int \sin[a+bx] f^{(n)}[x] dx$

**1:**  $\int \sin[a+bx] f^{(n)}[x] dx$  when  $n \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If  $n \in \mathbb{Z}^+$ , then

$$\int \sin[a+bx] f^{(n)}[x] dx \rightarrow \sin[a+bx] f^{(n-1)}[x] - b \int \cos[a+bx] f^{(n-1)}[x] dx$$

Program code:

```
Int[Sin[a_.+b_.*x_]*Derivative[n_][f_][x_],x_Symbol] :=
  Sin[a+b*x]*Derivative[n-1][f][x] - b*Int[Cos[a+b*x]*Derivative[n-1][f][x],x] /;
FreeQ[{a,b,f},x] && IGtQ[n,0]
```

```
Int[Cos[a_.*b_.*x_]*Derivative[n_][f_][x_],x_Symbol] :=
  Cos[a+b*x]*Derivative[n-1][f][x] + b*Int[Sin[a+b*x]*Derivative[n-1][f][x],x] /;
FreeQ[{a,b,f},x] && IGtQ[n,0]
```

**2:**  $\int \sin[a + b x] f^{(n)}[x] dx$  when  $n \in \mathbb{Z}^-$

Derivation: Integration by parts

Rule: If  $n \in \mathbb{Z}^-$ , then

$$\int \sin[a + b x] f^{(n)}[x] dx \rightarrow -\frac{\cos[a + b x] f^{(n)}[x]}{b} + \frac{1}{b} \int \cos[a + b x] f^{(n+1)}[x] dx$$

Program code:

```
Int[Sin[a_.*b_.*x_]*Derivative[n_][f_][x_],x_Symbol] :=
  -Cos[a+b*x]*Derivative[n][f][x]/b + 1/b*Int[Cos[a+b*x]*Derivative[n+1][f][x],x] /;
FreeQ[{a,b,f},x] && ILtQ[n,0]
```

```
Int[Cos[a_.*b_.*x_]*Derivative[n_][f_][x_],x_Symbol] :=
  Sin[a+b*x]*Derivative[n][f][x]/b - 1/b*Int[Sin[a+b*x]*Derivative[n+1][f][x],x] /;
FreeQ[{a,b,f},x] && ILtQ[n,0]
```

4:  $\int F[f^{(n-1)}[x]] f^{(n)}[x] dx$

Reference: G&R 2.02.7

Derivation: Integration by substitution

Basis:  $F[f[x]] f'[x] == \text{Subst}[F[x], x, f[x]] f'[x]$

Basis:  $F[f^{(n-1)}[x]] f^{(n)}[x] == \text{Subst}[F[x], x, f^{(n-1)}[x]] \partial_x f^{(n-1)}[x]$

Rule:

$$\int F[f^{(n-1)}[x]] f^{(n)}[x] dx \rightarrow \text{Subst}\left[\int F[x] dx, x, f^{(n-1)}[x]\right]$$

Program code:

```
Int[u_*Derivative[n_][f_][x_],x_Symbol] :=
  Subst[Int[SimplifyIntegrand[SubstFor[Derivative[n-1][f][x],u,x],x],x,Derivative[n-1][f][x]] /;
  FreeQ[{f,n},x] && FunctionOfQ[Derivative[n-1][f][x],u,x]
```

$$5: \int F[f^{(m-1)}[x] g^{(n-1)}[x]] (a f^{(m)}[x] g^{(n-1)}[x] + a f^{(m-1)}[x] g^{(n)}[x]) dx$$

Derivation: Integration by substitution

$$\text{Basis: } F[f[x] g[x]] (a f'[x] g[x] + a f[x] g'[x]) = a \text{Subst}[F[x], x, f[x] g[x]] \partial_x (f[x] g[x])$$

$$\text{Basis: } F[f^{(m-1)}[x] g^{(n-1)}[x]] (a f^{(m)}[x] g^{(n-1)}[x] + a f^{(m-1)}[x] g^{(n)}[x]) = \\ a \text{Subst}[F[x], x, f^{(m-1)}[x] g^{(n-1)}[x]] \partial_x (f^{(m-1)}[x] g^{(n-1)}[x])$$

Rule:

$$\int F[f^{(m-1)}[x] g^{(n-1)}[x]] (a f^{(m)}[x] g^{(n-1)}[x] + a f^{(m-1)}[x] g^{(n)}[x]) dx \rightarrow a \text{Subst}\left[\int F[x] dx, x, f^{(m-1)}[x] g^{(n-1)}[x]\right]$$

Program code:

```
Int[u_*(a_.*Derivative[1][f_][x_]*g_[x_]+a_.*f_[x_]*Derivative[1][g_][x_]),x_Symbol] :=
  a*Subst[Int[SimplifyIntegrand[SubstFor[f[x]*g[x],u,x],x],x,f[x]*g[x]] /;
  FreeQ[{a,f,g},x] && FunctionOfQ[f[x]*g[x],u,x]
```

```
Int[u_*(a_.*Derivative[m_][f_][x_]*g_[x_]+a_.*Derivative[m1_][f_][x_]*Derivative[1][g_][x_]),x_Symbol] :=
  a*Subst[Int[SimplifyIntegrand[SubstFor[Derivative[m-1][f][x]*g[x],u,x],x],x,Derivative[m-1][f][x]*g[x]] /;
  FreeQ[{a,f,g,m},x] && EqQ[m1,m-1] && FunctionOfQ[Derivative[m-1][f][x]*g[x],u,x]
```

```
Int[u_*(a_.*Derivative[m_][f_][x_]*Derivative[n1_][g_][x_]+a_.*Derivative[m1_][f_][x_]*Derivative[n_][g_][x_]),x_Symbol] :=
  a*Subst[Int[SimplifyIntegrand[SubstFor[Derivative[m-1][f][x]*Derivative[n-1][g][x],u,x],x],x,Derivative[m-1][f][x]*Derivative[n-1][g][x]] /;
  FreeQ[{a,f,g,m,n},x] && EqQ[m1,m-1] && EqQ[n1,n-1] && FunctionOfQ[Derivative[m-1][f][x]*Derivative[n-1][g][x],u,x]
```

**6:**  $\int F[f^{(m-1)}[x]^{p+1} g^{(n-1)}[x]] f^{(m-1)}[x]^p (a f^{(m)}[x] g^{(n-1)}[x] + b f^{(m-1)}[x] g^{(n)}[x]) dx$  when  $a = b(p+1)$

Derivation: Integration by substitution

Basis: If  $a = b(p+1)$ , then

$$F[f^{(m-1)}[x]^{p+1} g^{(n-1)}[x]] f^{(m-1)}[x]^p (a f^{(m)}[x] g^{(n-1)}[x] + b f^{(m-1)}[x] g^{(n)}[x]) = b \text{Subst}\left[F[x], x, f^{(m-1)}[x]^{p+1} g^{(n-1)}[x]\right] \partial_x (f^{(m-1)}[x]^{p+1} g^{(n-1)}[x])$$

Rule: If  $a = b(p+1)$ , then

$$\int F[f^{(m-1)}[x]^{p+1} g^{(n-1)}[x]] f^{(m-1)}[x]^p (a f^{(m)}[x] g^{(n-1)}[x] + b f^{(m-1)}[x] g^{(n)}[x]) dx \rightarrow b \text{Subst}\left[\int F[x] dx, x, f^{(m-1)}[x]^{p+1} g^{(n-1)}[x]\right]$$

Program code:

```
Int[u_*f_[x_]^p.*(a_.*Derivative[1][f_][x_]*g_[x_]+b_.*f_[x_]*Derivative[1][g_][x_]),x_Symbol] :=
  b*Subst[Int[SimplifyIntegrand[SubstFor[f[x]^(p+1)*g[x],u,x],x],x,f[x]^(p+1)*g[x]] /;
  FreeQ[{a,b,f,g,p},x] && EqQ[a,b*(p+1)] && FunctionOfQ[f[x]^(p+1)*g[x],u,x]
```

```
Int[u_*Derivative[m1_][f_][x_]^p.*(
  (a_.*Derivative[m_][f_][x_]*g_[x_]+b_.*Derivative[m1_][f_][x_]*Derivative[1][g_][x_]),x_Symbol] :=
  b*Subst[Int[SimplifyIntegrand[SubstFor[Derivative[m-1][f][x]^(p+1)*g[x],u,x],x],x,
  Derivative[m-1][f][x]^(p+1)*g[x]] /;
  FreeQ[{a,b,f,g,m,p},x] && EqQ[m1,m-1] && EqQ[a,b*(p+1)] && FunctionOfQ[Derivative[m-1][f][x]^(p+1)*g[x],u,x]
```

```
Int[u_*g_[x_]^q.*(
  (a_.*Derivative[m_][f_][x_]*g_[x_]+b_.*Derivative[m1_][f_][x_]*Derivative[1][g_][x_]),x_Symbol] :=
  a*Subst[Int[SimplifyIntegrand[SubstFor[Derivative[m-1][f][x]*g[x]^(q+1),u,x],x],x,
  Derivative[m-1][f][x]*g[x]^(q+1)] /;
  FreeQ[{a,b,f,g,m,q},x] && EqQ[m1,m-1] && EqQ[a*(q+1),b] && FunctionOfQ[Derivative[m-1][f][x]*g[x]^(q+1),u,x]
```

```

Int[u_*Derivative[m1_][f_][x_]^p_.*
  (a_.*Derivative[m_][f_][x_] * Derivative[n1_][g_][x_] + b_.*Derivative[m1_][f_][x_] * Derivative[n_][g_][x_]), x_Symbol] :=
  b*Subst[Int[SimplifyIntegrand[SubstFor[Derivative[m-1][f][x]^(p+1)*Derivative[n-1][g][x], u, x], x], x], x,
    Derivative[m-1][f][x]^(p+1)*Derivative[n-1][g][x]] /;
FreeQ[{a, b, f, g, m, n, p}, x] && EqQ[m1, m-1] && EqQ[n1, n-1] && EqQ[a, b*(p+1)] &&
FunctionOfQ[Derivative[m-1][f][x]^(p+1)*Derivative[n-1][g][x], u, x]

```

**7:**  $\int F[f^{(m-1)}[x]^{p+1} g^{(n-1)}[x]^{q+1}] f^{(m-1)}[x]^p g^{(n-1)}[x]^q (a f^{(m)}[x] g^{(n-1)}[x] + b f^{(m-1)}[x] g^{(n)}[x]) dx$  when  $a(q+1) = b(p+1)$

Derivation: Integration by substitution

Basis: If  $a(q+1) = b(p+1)$ , then  $F[f[x]^{p+1} g[x]^{q+1}] f[x]^p g[x]^q (a f'[x] g[x] + b f[x] g'[x]) =$   
 $\frac{a}{p+1} \text{Subst}[F[x], x, f[x]^{p+1} g[x]^{q+1}] \partial_x (f[x]^{p+1} g[x]^{q+1})$

Basis: If  $a(q+1) = b(p+1)$ , then

$$F[f^{(m-1)}[x]^{p+1} g^{(n-1)}[x]^{q+1}] f^{(m-1)}[x]^p g^{(n-1)}[x]^q (a f^{(m)}[x] g^{(n-1)}[x] + b f^{(m-1)}[x] g^{(n)}[x]) =$$

$$\frac{a}{p+1} \text{Subst}[F[x], x, f^{(m-1)}[x]^{p+1} g^{(n-1)}[x]^{q+1}] \partial_x (f^{(m-1)}[x]^{p+1} g^{(n-1)}[x]^{q+1})$$

Rule: If  $a(q+1) = b(p+1)$ , then

$$\int F[f^{(m-1)}[x]^{p+1} g^{(n-1)}[x]^{q+1}] f^{(m-1)}[x]^p g^{(n-1)}[x]^q (a f^{(m)}[x] g^{(n-1)}[x] + b f^{(m-1)}[x] g^{(n)}[x]) dx \rightarrow$$

$$\frac{a}{p+1} \text{Subst}\left[\int F[x] dx, x, f^{(m-1)}[x]^{p+1} g^{(n-1)}[x]^{q+1}\right]$$

Program code:

```

Int[u_*f[x_]^p_.*g[x_]^q_.*(a_.*Derivative[1][f_][x_]*g[x_]+b_.*f[x_]*Derivative[1][g_][x_]), x_Symbol] :=
  a/(p+1)*Subst[Int[SimplifyIntegrand[SubstFor[f[x]^(p+1)*g[x]^(q+1), u, x], x], x], x, f[x]^(p+1)*g[x]^(q+1)] /;
FreeQ[{a, b, f, g, p, q}, x] && EqQ[a*(q+1), b*(p+1)] && FunctionOfQ[f[x]^(p+1)*g[x]^(q+1), u, x]

```

```

Int[u_*Derivative[m1_][f_][x_]^p_.*g_[x_]^q_.*
  (a_.*Derivative[m_][f_][x_]g_[x_]+b_.*Derivative[m1_][f_][x_]
  *Derivative[1][g_][x_]),x_Symbol] :=
a/(p+1)*Subst[Int[SimplifyIntegrand[SubstFor[Derivative[m-1][f][x]^(p+1)*g[x]^(q+1),u,x],x],x,
  Derivative[m-1][f][x]^(p+1)*g[x]^(q+1)] /;
FreeQ[{a,b,f,g,m,p,q},x] && EqQ[m1,m-1] && EqQ[a*(q+1),b*(p+1)] && FunctionOfQ[Derivative[m-1][f][x]^(p+1)*g[x]^(q+1),u,x]

```

```

Int[u_*Derivative[m1_][f_][x_]^p_.*Derivative[n1_][g_][x_]^q_.*
  (a_.*Derivative[m_][f_][x_]g_[x_]+b_.*Derivative[m1_][f_][x_]
  *Derivative[n_][g_][x_]),x_Symbol] :=
a/(p+1)*Subst[Int[SimplifyIntegrand[SubstFor[Derivative[m-1][f][x]^(p+1)*Derivative[n-1][g][x]^(q+1),u,x],x],x,
  Derivative[m-1][f][x]^(p+1)*Derivative[n-1][g][x]^(q+1)] /;
FreeQ[{a,b,f,g,m,n,p,q},x] && EqQ[m1,m-1] && EqQ[n1,n-1] && EqQ[a*(q+1),b*(p+1)] &&
FunctionOfQ[Derivative[m-1][f][x]^(p+1)*Derivative[n-1][g][x]^(q+1),u,x]

```

2:  $\int f'[x] g[x] + f[x] g'[x] dx$

Reference: G&R 2.02.5

Derivation: Inverse of derivative of a product rule

Rule:

$$\int f'[x] g[x] + f[x] g'[x] dx \rightarrow f[x] g[x]$$

Program code:

```

Int[f_'[x_]*g_[x_] + f_[x_]*g_'[x_],x_Symbol] :=
  f[x]*g[x] /;
FreeQ[{f,g},x]

```



3:  $\int \frac{f'[x] g[x] - f[x] g'[x]}{g[x]^2} dx$

Reference: G&R 2.02.11

Derivation: Inverse of derivative of a quotient rule

Rule:

$$\int \frac{f'[x] g[x] - f[x] g'[x]}{g[x]^2} dx \rightarrow \frac{f[x]}{g[x]}$$

Program code:

```
Int[(f'[x]*g[x] - f[x]*g'[x])/g[x]^2,x_Symbol] :=
  f[x]/g[x] /;
FreeQ[{f,g},x]
```

4: 
$$\int \frac{f'[x] g[x] - f[x] g'[x]}{f[x] g[x]} dx$$

Reference: G&R 2.02.12

Derivation: Inverse of derivative of log of a quotient rule

Rule:

$$\int \frac{f'[x] g[x] - f[x] g'[x]}{f[x] g[x]} dx \rightarrow \text{Log}\left[\frac{f[x]}{g[x]}\right]$$

Program code:

```
Int[(f'[x_]*g_[x_] - f_[x_]*g_'[x_])/(f_[x_]*g_[x_]),x_Symbol] :=
  Log[f[x]/g[x]] /;
FreeQ[{f,g},x]
```