Rules for integrands of the form $F^{c\,(a+b\,x)}\,\text{Trig}[d+e\,x]^n$

1. $\int F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]^n\,dx$

   1. $\int F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]^n\,dx$ when $e^2\,n^2+b^2\,c^2\,\text{Log}[F]^2\neq 0\land n>0$

      1: $\int F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]\,dx$ when $e^2+b^2\,c^2\,\text{Log}[F]^2\neq 0$

Reference: CRC 533, A&S 4.3.136

Reference: CRC 538, A&S 4.3.137

Rule: If $e^2+b^2\,c^2\,\text{Log}[F]^2\neq 0$, then

$$\int F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]\,dx\,\longrightarrow\,\frac{b\,c\,\text{Log}[F]\,F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]}{e^2+b^2\,c^2\,\text{Log}[F]^2}-\frac{e\,F^{c\,(a+b\,x)}\,\text{Cos}[d+e\,x]}{e^2+b^2\,c^2\,\text{Log}[F]^2}$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sin[d_.+e_.*x_],x_Symbol] :=
  b*c*Log[F]*F^(c*(a+b*x))*Sin[d+e*x]/(e^2+b^2*c^2*Log[F]^2) -
  e*F^(c*(a+b*x))*Cos[d+e*x]/(e^2+b^2*c^2*Log[F]^2) /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2+b^2*c^2*Log[F]^2,0]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Cos[d_.+e_.*x_],x_Symbol] :=
  b*c*Log[F]*F^(c*(a+b*x))*Cos[d+e*x]/(e^2+b^2*c^2*Log[F]^2) +
  e*F^(c*(a+b*x))*Sin[d+e*x]/(e^2+b^2*c^2*Log[F]^2) /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2+b^2*c^2*Log[F]^2,0]
```

2: $\int F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]^n\,dx$ when $e^2\,n^2+b^2\,c^2\,\text{Log}[F]^2\neq 0\,\land\,n>1$

Reference: CRC 542, A&S 4.3.138

Reference: CRC 543, A&S 4.3.139

Rule: If $e^2\,n^2+b^2\,c^2\,\text{Log}[F]^2\neq 0\,\land\,n>1$, then

$$\int F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]^n\,dx\,\longrightarrow$$

$$\frac{b\,c\,\text{Log}[F]\,F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]^n}{e^2\,n^2+b^2\,c^2\,\text{Log}[F]^2}-\frac{e\,n\,F^{c\,(a+b\,x)}\,\text{Cos}[d+e\,x]\,\text{Sin}[d+e\,x]^{n-1}}{e^2\,n^2+b^2\,c^2\,\text{Log}[F]^2}+\frac{n\,(n-1)\,e^2}{e^2\,n^2+b^2\,c^2\,\text{Log}[F]^2}\int F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]^{n-2}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sin[d_.+e_.*x_]^n_,x_Symbol] :=
  b*c*Log[F]*F^(c*(a+b*x))*Sin[d+e*x]^n/(e^2*n^2+b^2*c^2*Log[F]^2) -
  e*n*F^(c*(a+b*x))*Cos[d+e*x]*Sin[d+e*x]^(n-1)/(e^2*n^2+b^2*c^2*Log[F]^2) +
  (n*(n-1)*e^2)/(e^2*n^2+b^2*c^2*Log[F]^2)*Int[F^(c*(a+b*x))*Sin[d+e*x]^(n-2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*n^2+b^2*c^2*Log[F]^2,0] && GtQ[n,1]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Cos[d_.+e_.*x_]^m_,x_Symbol] :=
  b*c*Log[F]*F^(c*(a+b*x))*Cos[d+e*x]^m/(e^2*m^2+b^2*c^2*Log[F]^2) +
  e*m*F^(c*(a+b*x))*Sin[d+e*x]*Cos[d+e*x]^(m-1)/(e^2*m^2+b^2*c^2*Log[F]^2) +
  (m*(m-1)*e^2)/(e^2*m^2+b^2*c^2*Log[F]^2)*Int[F^(c*(a+b*x))*Cos[d+e*x]^(m-2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*m^2+b^2*c^2*Log[F]^2,0] && GtQ[m,1]
```

2: $\int F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]^n\,dx$ when $e^2\,(n+2)^2+b^2\,c^2\,\mathrm{Log}[F]^2==0\,\land\,n\neq-1\,\land\,n\neq-2$

Reference: CRC 551 when $e^2\,(n+2)^2+b^2\,c^2\,\mathrm{Log}[F]^2==0$

Reference: CRC 552 when $e^2\,(n+2)^2+b^2\,c^2\,\mathrm{Log}[F]^2==0$

Rule: If $e^2\,(n+2)^2+b^2\,c^2\,\mathrm{Log}[F]^2==0\,\land\,n\neq-1\,\land\,n\neq-2$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]^n\,dx\,\longrightarrow\,-\frac{b\,c\,\mathrm{Log}[F]\,F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]^{n+2}}{e^2\,(n+1)\,(n+2)}+\frac{F^{c\,(a+b\,x)}\,\mathrm{Cos}[d+e\,x]\,\mathrm{Sin}[d+e\,x]^{n+1}}{e\,(n+1)}$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sin[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Sin[d+e*x]^(n+2)/(e^2*(n+1)*(n+2)) +
  F^(c*(a+b*x))*Cos[d+e*x]*Sin[d+e*x]^(n+1)/(e*(n+1)) /;
FreeQ[{F,a,b,c,d,e,n},x] && EqQ[e^2*(n+2)^2+b^2*c^2*Log[F]^2,0] && NeQ[n,-1] && NeQ[n,-2]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Cos[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Cos[d+e*x]^(n+2)/(e^2*(n+1)*(n+2)) -
  F^(c*(a+b*x))*Sin[d+e*x]*Cos[d+e*x]^(n+1)/(e*(n+1)) /;
FreeQ[{F,a,b,c,d,e,n},x] && EqQ[e^2*(n+2)^2+b^2*c^2*Log[F]^2,0] && NeQ[n,-1] && NeQ[n,-2]
```

3: $\int F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]^n\,dx$ when $e^2\,(n+2)^2+b^2\,c^2\,\mathrm{Log}[F]^2\neq0\,\land\,n<-1\,\land\,n\neq-2$

Reference: CRC 551, CRC 542 inverted

Reference: CRC 552, CRC 543 inverted

Rule: If $e^2\,(n+2)^2+b^2\,c^2\,\mathrm{Log}[F]^2\neq0\,\land\,n<-1\,\land\,n\neq-2$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]^n\,dx\,\longrightarrow$$

$$-\frac{b\,c\,\mathrm{Log}[F]\,F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]^{n+2}}{e^2\,(n+1)\,(n+2)}+\frac{F^{c\,(a+b\,x)}\,\mathrm{Cos}[d+e\,x]\,\mathrm{Sin}[d+e\,x]^{n+1}}{e\,(n+1)}+\frac{e^2\,(n+2)^2+b^2\,c^2\,\mathrm{Log}[F]^2}{e^2\,(n+1)\,(n+2)}\int F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]^{n+2}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sin[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Sin[d+e*x]^(n+2)/(e^2*(n+1)*(n+2)) +
  F^(c*(a+b*x))*Cos[d+e*x]*Sin[d+e*x]^(n+1)/(e*(n+1)) +
  (e^2*(n+2)^2+b^2*c^2*Log[F]^2)/(e^2*(n+1)*(n+2))*Int[F^(c*(a+b*x))*Sin[d+e*x]^(n+2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*(n+2)^2+b^2*c^2*Log[F]^2,0] && LtQ[n,-1] && NeQ[n,-2]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Cos[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Cos[d+e*x]^(n+2)/(e^2*(n+1)*(n+2)) -
  F^(c*(a+b*x))*Sin[d+e*x]*Cos[d+e*x]^(n+1)/(e*(n+1)) +
  (e^2*(n+2)^2+b^2*c^2*Log[F]^2)/(e^2*(n+1)*(n+2))*Int[F^(c*(a+b*x))*Cos[d+e*x]^(n+2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*(n+2)^2+b^2*c^2*Log[F]^2,0] && LtQ[n,-1] && NeQ[n,-2]
```

4: $\int F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]^n\,dx$ when $n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\mathrm{Sin}[z] = -\frac{1}{2}\,i\,e^{-i\,z}\left(-1+e^{2\,i\,z}\right)$

Basis: $\partial_x \frac{e^{i\,n\,(d+e\,x)}\,\mathrm{Sin}[d+e\,x]^n}{\left(-1+e^{2\,i\,(d+e\,x)}\right)^n} = 0$

Rule: If $n\notin\mathbb{Z}$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]^n\,dx \to \frac{e^{i\,n\,(d+e\,x)}\,\mathrm{Sin}[d+e\,x]^n}{\left(-1+e^{2\,i\,(d+e\,x)}\right)^n}\int F^{c\,(a+b\,x)}\,\frac{\left(-1+e^{2\,i\,(d+e\,x)}\right)^n}{e^{i\,n\,(d+e\,x)}}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sin[d_.+e_.*x_]^n_,x_Symbol] :=
  E^(I*n*(d+e*x))*Sin[d+e*x]^n/(-1+E^(2*I*(d+e*x)))^n*Int[F^(c*(a+b*x))*(-1+E^(2*I*(d+e*x)))^n/E^(I*n*(d+e*x)),x] /;
FreeQ[{F,a,b,c,d,e,n},x] && Not[IntegerQ[n]]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Cos[d_.+e_.*x_]^n_,x_Symbol] :=
  E^(I*n*(d+e*x))*Cos[d+e*x]^n/(1+E^(2*I*(d+e*x)))^n*Int[F^(c*(a+b*x))*(1+E^(2*I*(d+e*x)))^n/E^(I*n*(d+e*x)),x] /;
FreeQ[{F,a,b,c,d,e,n},x] && Not[IntegerQ[n]]
```

2: $\int F^{c\,(a+b\,x)}\,\mathrm{Tan}[d+e\,x]^n\,dx$ when $n\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: If $n\in\mathbb{Z}$, then $\mathrm{Tan}[z]^n = i^n\,\frac{\left(1-e^{2\,i\,z}\right)^n}{\left(1+e^{2\,i\,z}\right)^n}$

Rule: If $n\in\mathbb{Z}$, then

$$\int F^{c\,(a+b\,x)}\,\operatorname{Tan}[d+e\,x]^n\,dx\;\longrightarrow\;i^n\int F^{c\,(a+b\,x)}\,\frac{\left(1-e^{2\,i\,(d+e\,x)}\right)^n}{\left(1+e^{2\,i\,(d+e\,x)}\right)^n}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Tan[d_.+e_.*x_]^n_.,x_Symbol] :=
  I^n*Int[ExpandIntegrand[F^(c*(a+b*x))*(1-E^(2*I*(d+e*x)))^n/(1+E^(2*I*(d+e*x)))^n,x],x] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[n]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Cot[d_.+e_.*x_]^n_.,x_Symbol] :=
  (-I)^n*Int[ExpandIntegrand[F^(c*(a+b*x))*(1+E^(2*I*(d+e*x)))^n/(1-E^(2*I*(d+e*x)))^n,x],x] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[n]
```

3. $\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx$

1: $\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx$ when $e^2\,n^2+b^2\,c^2\,\mathrm{Log}[F]^2\neq 0\,\wedge\,n<-1$

Reference: CRC 552 inverted

Reference: CRC 551 inverted

Rule: If $e^2\,n^2+b^2\,c^2\,\mathrm{Log}[F]^2\neq 0\,\wedge\,n<-1$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx\,\longrightarrow$$
$$\frac{b\,c\,\mathrm{Log}[F]\,F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n}{e^2\,n^2+b^2\,c^2\,\mathrm{Log}[F]^2}-\frac{e\,n\,F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^{n+1}\,\mathrm{Sin}[d+e\,x]}{e^2\,n^2+b^2\,c^2\,\mathrm{Log}[F]^2}+\frac{e^2\,n\,(n+1)}{e^2\,n^2+b^2\,c^2\,\mathrm{Log}[F]^2}\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^{n+2}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sec[d_.+e_.*x_]^n_,x_Symbol] :=
  b*c*Log[F]*F^(c*(a+b*x))*(Sec[d+e x]^n/(e^2*n^2+b^2*c^2*Log[F]^2)) -
   e*n*F^(c*(a+b*x))*Sec[d+e x]^(n+1)*(Sin[d+e x]/(e^2*n^2+b^2*c^2*Log[F]^2)) +
  e^2*n*((n+1)/(e^2*n^2+b^2*c^2*Log[F]^2))*Int[F^(c*(a+b*x))*Sec[d+e x]^(n+2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*n^2+b^2*c^2*Log[F]^2,0] && LtQ[n,-1]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Csc[d_.+e_.*x_]^n_,x_Symbol] :=
  b*c*Log[F]*F^(c*(a+b*x))*(Csc[d+e x]^n/(e^2*n^2+b^2*c^2*Log[F]^2)) +
   e*n*F^(c*(a+b*x))*Csc[d+e x]^(n+1)*(Cos[d+e x]/(e^2*n^2+b^2*c^2*Log[F]^2)) +
  e^2*n*((n+1)/(e^2*n^2+b^2*c^2*Log[F]^2))*Int[F^(c*(a+b*x))*Csc[d+e x]^(n+2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[e^2*n^2+b^2*c^2*Log[F]^2,0] && LtQ[n,-1]
```

2: $\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx$ when $e^2\,(n-2)^2+b^2\,c^2\,\mathrm{Log}[F]^2==0\,\land\,n\neq 1\,\land\,n\neq 2$

Reference: CRC 552 with $e^2\,(n-2)^2+b^2\,c^2\,\mathrm{Log}[F]^2==0$

Reference: CRC 551 with $e^2\,(n-2)^2+b^2\,c^2\,\mathrm{Log}[F]^2==0$

Rule: If $e^2\,(n-2)^2+b^2\,c^2\,\mathrm{Log}[F]^2==0\,\land\,n\neq 1\,\land\,n\neq 2$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx\,\longrightarrow\,-\frac{b\,c\,\mathrm{Log}[F]\,F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^{n-2}}{e^2\,(n-1)\,(n-2)}+\frac{F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^{n-1}\,\mathrm{Sin}[d+e\,x]}{e\,(n-1)}$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sec[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Sec[d+e x]^(n-2)/(e^2*(n-1)*(n-2)) +
  F^(c*(a+b*x))*Sec[d+e x]^(n-1)*Sin[d+e x]/(e*(n-1)) /;
FreeQ[{F,a,b,c,d,e,n},x] && EqQ[b^2*c^2*Log[F]^2+e^2*(n-2)^2,0] && NeQ[n,1] && NeQ[n,2]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Csc[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Csc[d+e x]^(n-2)/(e^2*(n-1)*(n-2)) +
  F^(c*(a+b*x))*Csc[d+e x]^(n-1)*Cos[d+e x]/(e*(n-1)) /;
FreeQ[{F,a,b,c,d,e,n},x] && EqQ[b^2*c^2*Log[F]^2+e^2*(n-2)^2,0] && NeQ[n,1] && NeQ[n,2]
```

3: $\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx$ when $e^2\,(n-2)^2+b^2\,c^2\,\mathrm{Log}[F]^2\neq 0\,\land\,n>1\,\land\,n\neq 2$

Reference: CRC 552

Reference: CRC 551

Rule: If $e^2\,(n-2)^2+b^2\,c^2\,\mathrm{Log}[F]^2\neq 0\,\land\,n>1\,\land\,n\neq 2$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx\,\longrightarrow$$

$$-\frac{b\,c\,\text{Log}[F]\,F^{c\,(a+b\,x)}\,\text{Sec}[d+e\,x]^{n-2}}{e^2\,(n-1)\,(n-2)}+\frac{F^{c\,(a+b\,x)}\,\text{Sec}[d+e\,x]^{n-1}\,\text{Sin}[d+e\,x]}{e\,(n-1)}+\frac{e^2\,(n-2)^2+b^2\,c^2\,\text{Log}[F]^2}{e^2\,(n-1)\,(n-2)}\int F^{c\,(a+b\,x)}\,\text{Sec}[d+e\,x]^{n-2}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sec[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Sec[d+e x]^(n-2)/(e^2*(n-1)*(n-2)) +
  F^(c*(a+b*x))*Sec[d+e x]^(n-1)*Sin[d+e x]/(e*(n-1)) +
  (e^2*(n-2)^2+b^2*c^2*Log[F]^2)/(e^2*(n-1)*(n-2))*Int[F^(c*(a+b*x))*Sec[d+e x]^(n-2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[b^2*c^2*Log[F]^2+e^2*(n-2)^2,0] && GtQ[n,1] && NeQ[n,2]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Csc[d_.+e_.*x_]^n_,x_Symbol] :=
  -b*c*Log[F]*F^(c*(a+b*x))*Csc[d+e x]^(n-2)/(e^2*(n-1)*(n-2)) -
  F^(c*(a+b*x))*Csc[d+e x]^(n-1)*Cos[d+e x]/(e*(n-1)) +
  (e^2*(n-2)^2+b^2*c^2*Log[F]^2)/(e^2*(n-1)*(n-2))*Int[F^(c*(a+b*x))*Csc[d+e x]^(n-2),x] /;
FreeQ[{F,a,b,c,d,e},x] && NeQ[b^2*c^2*Log[F]^2+e^2*(n-2)^2,0] && GtQ[n,1] && NeQ[n,2]
```

x: $\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx$ when $n\in\mathbb{Z}$

Derivation: Algebraic expansion

Basis: $\mathrm{Sec}[z]==\frac{2\,e^{i\,z}}{1+e^{2\,i\,z}}$

Basis: $\mathrm{Csc}[z]==\frac{2\,i\,e^{-i\,z}}{1-e^{-2\,i\,z}}$

Rule: If $n\in\mathbb{Z}$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx\,\longrightarrow\,2^n\int F^{c\,(a+b\,x)}\,\frac{e^{i\,n\,(d+e\,x)}}{\left(1+e^{2\,i\,(d+e\,x)}\right)^n}\,dx$$

Program code:

```
(* Int[F_^(c_.*(a_.+b_.*x_))*Sec[d_.+e_.*x_]^n_.,x_Symbol] :=
  2^n*Int[SimplifyIntegrand[F^(c*(a+b*x))*E^(I*n*(d+e*x))/(1+E^(2*I*(d+e*x)))^n,x],x] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[n] *)
```

```
(* Int[F_^(c_.*(a_.+b_.*x_))*Csc[d_.+e_.*x_]^n_.,x_Symbol] :=
  (2*I)^n*Int[SimplifyIntegrand[F^(c*(a+b*x))*E^(-I*n*(d+e*x))/(1-E^(-2*I*(d+e*x)))^n,x],x] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[n] *)
```

4: $\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx$ when $n \in \mathbb{Z}$

Rule: If $n \in \mathbb{Z}$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx \to \frac{2^n\,e^{i\,n\,(d+e\,x)}\,F^{c\,(a+b\,x)}}{i\,e\,n+b\,c\,\mathrm{Log}[F]}\,\mathrm{Hypergeometric2F1}\left[n,\,\frac{n}{2}-\frac{i\,b\,c\,\mathrm{Log}[F]}{2\,e},\,1+\frac{n}{2}-\frac{i\,b\,c\,\mathrm{Log}[F]}{2\,e},\,-e^{2\,i\,(d+e\,x)}\right]$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sec[d_.+k_.*Pi+e_.*x_]^n_.,x_Symbol] :=
  2^n*E^(I*k*n*Pi)*E^(I*n*(d+e*x))*F^(c*(a+b*x))/(I*e*n+b*c*Log[F])*
    Hypergeometric2F1[n,n/2-I*b*c*Log[F]/(2*e),1+n/2-I*b*c*Log[F]/(2*e),-E^(2*I*k*Pi)*E^(2*I*(d+e*x))] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[4*k] && IntegerQ[n]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Sec[d_.+e_.*x_]^n_.,x_Symbol] :=
  2^n*E^(I*n*(d+e*x))*F^(c*(a+b*x))/(I*e*n+b*c*Log[F])*
    Hypergeometric2F1[n,n/2-I*b*c*Log[F]/(2*e),1+n/2-I*b*c*Log[F]/(2*e),-E^(2*I*(d+e*x))] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[n]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Csc[d_.+k_.*Pi+e_.*x_]^n_.,x_Symbol] :=
  (-2*I)^n*E^(I*k*n*Pi)*E^(I*n*(d+e*x))*(F^(c*(a+b*x))/(I*e*n+b*c*Log[F]))*
    Hypergeometric2F1[n,n/2-I*b*c*Log[F]/(2*e),1+n/2-I*b*c*Log[F]/(2*e),E^(2*I*k*Pi)*E^(2*I*(d+e*x))] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[4*k] && IntegerQ[n]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Csc[d_.+e_.*x_]^n_.,x_Symbol] :=
  (-2*I)^n*E^(I*n*(d+e*x))*(F^(c*(a+b*x))/(I*e*n+b*c*Log[F]))*
    Hypergeometric2F1[n,n/2-I*b*c*Log[F]/(2*e),1+n/2-I*b*c*Log[F]/(2*e),E^(2*I*(d+e*x))] /;
FreeQ[{F,a,b,c,d,e},x] && IntegerQ[n]
```

5: $\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx$ when $n\notin\mathbb{Z}$

Derivation: Piecewise constant extraction

Basis: $\partial_x \frac{\left(1+e^{2\,i\,(d+e\,x)}\right)^n\,\mathrm{Sec}[d+e\,x]^n}{e^{i\,n\,(d+e\,x)}} == 0$

Rule: If $n\notin\mathbb{Z}$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Sec}[d+e\,x]^n\,dx \longrightarrow \frac{\left(1+e^{2\,i\,(d+e\,x)}\right)^n\,\mathrm{Sec}[d+e\,x]^n}{e^{i\,n\,(d+e\,x)}}\int F^{c\,(a+b\,x)}\,\frac{e^{i\,n\,(d+e\,x)}}{\left(1+e^{2\,i\,(d+e\,x)}\right)^n}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sec[d_.+e_.*x_]^n_.,x_Symbol] :=
  (1+E^(2*I*(d+e*x)))^n*Sec[d+e*x]^n/E^(I*n*(d+e*x))*Int[SimplifyIntegrand[F^(c*(a+b*x))*E^(I*n*(d+e*x))/(1+E^(2*I*(d+e*x)))^n,x],x] /
FreeQ[{F,a,b,c,d,e},x] && Not[IntegerQ[n]]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Csc[d_.+e_.*x_]^n_.,x_Symbol] :=
  (1-E^(-2*I*(d+e*x)))^n*Csc[d+e*x]^n/E^(-I*n*(d+e*x))*Int[SimplifyIntegrand[F^(c*(a+b*x))*E^(-I*n*(d+e*x))/(1-E^(-2*I*(d+e*x)))^n,x],
FreeQ[{F,a,b,c,d,e},x] && Not[IntegerQ[n]]
```

4. $\int u\, F^{c\,(a+b\,x)}\,(f+g\,\text{Sin}[d+e\,x])^n\,dx$ when $f^2-g^2=0$

1: $\int F^{c\,(a+b\,x)}\,(f+g\,\text{Sin}[d+e\,x])^n\,dx$ when $f^2-g^2=0\,\wedge\,n\in\mathbb{Z}$

Derivation: Algebraic simplification

Basis: If $f^2-g^2=0$, then $f+g\,\text{Sin}[z]=2\,f\,\text{Cos}\left[\frac{z}{2}-\frac{f\,\pi}{4\,g}\right]^2$

Basis: If $f-g=0$, then $f+g\,\text{Cos}[z]=2\,f\,\text{Cos}\left[\frac{z}{2}\right]^2$

Basis: If $f+g=0$, then $f+g\,\text{Cos}[z]=2\,f\,\text{Sin}\left[\frac{z}{2}\right]^2$

Rule: If $f^2-g^2=0\,\wedge\,n\in\mathbb{Z}$, then

$$\int F^{c\,(a+b\,x)}\,(f+g\,\text{Sin}[d+e\,x])^n\,dx\,\longrightarrow\,2^n\,f^n\int F^{c\,(a+b\,x)}\,\text{Cos}\left[\frac{d}{2}+\frac{e\,x}{2}-\frac{f\,\pi}{4\,g}\right]^{2\,n}dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*(f_+g_.*Sin[d_.+e_.*x_])^n_.,x_Symbol] :=
  2^n*f^n*Int[F^(c*(a+b*x))*Cos[d/2+e*x/2-f*Pi/(4*g)]^(2*n),x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && EqQ[f^2-g^2,0] && ILtQ[n,0]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*(f_+g_.*Cos[d_.+e_.*x_])^n_.,x_Symbol] :=
  2^n*f^n*Int[F^(c*(a+b*x))*Cos[d/2+e*x/2]^(2*n),x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && EqQ[f-g,0] && ILtQ[n,0]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*(f_+g_.*Cos[d_.+e_.*x_])^n_.,x_Symbol] :=
  2^n*f^n*Int[F^(c*(a+b*x))*Sin[d/2+e*x/2]^(2*n),x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && EqQ[f+g,0] && ILtQ[n,0]
```

2: $\int F^{c\,(a+b\,x)}\,\mathrm{Cos}[d+e\,x]^m\,(f+g\,\mathrm{Sin}[d+e\,x])^n\,dx$ when $f^2-g^2==0\,\land\,(m\mid n)\in\mathbb{Z}\,\land\,m+n==0$

Derivation: Algebraic simplification

Basis: If $f^2-g^2==0$, then $\frac{\mathrm{Cos}[z]}{f+g\,\mathrm{Sin}[z]}==\frac{1}{g}\,\mathrm{Tan}\left[\frac{f\,\pi}{4\,g}-\frac{z}{2}\right]$

Basis: If $f-g==0$, then $\frac{\mathrm{Sin}[z]}{f+g\,\mathrm{Cos}[z]}==\frac{1}{f}\,\mathrm{Tan}\left[\frac{z}{2}\right]$

Basis: If $f+g==0$, then $\frac{\mathrm{Sin}[z]}{f+g\,\mathrm{Cos}[z]}==\frac{1}{f}\,\mathrm{Cot}\left[\frac{z}{2}\right]$

Rule: If $f^2-g^2==0\,\land\,(m\mid n)\in\mathbb{Z}\,\land\,m+n==0$, then

$$\int F^{c\,(a+b\,x)}\,\mathrm{Cos}[d+e\,x]^m\,(f+g\,\mathrm{Sin}[d+e\,x])^n\,dx\;\longrightarrow\;g^n\int F^{c\,(a+b\,x)}\,\mathrm{Tan}\left[\frac{f\,\pi}{4\,g}-\frac{d}{2}-\frac{e\,x}{2}\right]^m\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Cos[d_.+e_.*x_]^m_.*(f_+g_.*Sin[d_.+e_.*x_])^n_.,x_Symbol] :=
  g^n*Int[F^(c*(a+b*x))*Tan[f*Pi/(4*g)-d/2-e*x/2]^m,x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && EqQ[f^2-g^2,0] && IntegersQ[m,n] && EqQ[m+n,0]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Sin[d_.+e_.*x_]^m_.*(f_+g_.*Cos[d_.+e_.*x_])^n_.,x_Symbol] :=
  f^n*Int[F^(c*(a+b*x))*Tan[d/2+e*x/2]^m,x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && EqQ[f-g,0] && IntegersQ[m,n] && EqQ[m+n,0]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*Sin[d_.+e_.*x_]^m_.*(f_+g_.*Cos[d_.+e_.*x_])^n_.,x_Symbol] :=
  f^n*Int[F^(c*(a+b*x))*Cot[d/2+e*x/2]^m,x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && EqQ[f+g,0] && IntegersQ[m,n] && EqQ[m+n,0]
```

3: $\int F^{c\,(a+b\,x)}\,\frac{h+i\,\mathrm{Cos}[d+e\,x]}{f+g\,\mathrm{Sin}[d+e\,x]}\,dx$ when $f^2-g^2==0\,\wedge\,h^2-i^2==0\,\wedge\,g\,h+f\,i==0$

Derivation: Algebraic simplification

Basis: $\frac{h+i\,\mathrm{Cos}[z]}{f+g\,\mathrm{Sin}[z]}==\frac{2\,i\,\mathrm{Cos}[z]}{f+g\,\mathrm{Sin}[z]}+\frac{h-i\,\mathrm{Cos}[z]}{f+g\,\mathrm{Sin}[z]}$

Rule: If $f^2-g^2==0\,\wedge\,h^2-i^2==0\,\wedge\,g\,h+f\,i==0$, then

$$\int F^{c\,(a+b\,x)}\,\frac{h+i\,\mathrm{Cos}[d+e\,x]}{f+g\,\mathrm{Sin}[d+e\,x]}\,dx\;\longrightarrow\;2\,i\int F^{c\,(a+b\,x)}\,\frac{\mathrm{Cos}[d+e\,x]}{f+g\,\mathrm{Sin}[d+e\,x]}\,dx+\int F^{c\,(a+b\,x)}\,\frac{h-i\,\mathrm{Cos}[d+e\,x]}{f+g\,\mathrm{Sin}[d+e\,x]}\,dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*(h_+i_.*Cos[d_.+e_.*x_])/(f_+g_.*Sin[d_.+e_.*x_]),x_Symbol] :=
  2*i*Int[F^(c*(a+b*x))*(Cos[d+e*x]/(f+g*Sin[d+e*x])),x] +
  Int[F^(c*(a+b*x))*((h-i*Cos[d+e*x])/(f+g*Sin[d+e*x])),x] /;
FreeQ[{F,a,b,c,d,e,f,g,h,i},x] && EqQ[f^2-g^2,0] && EqQ[h^2-i^2,0] && EqQ[g*h-f*i,0]
```

```
Int[F_^(c_.*(a_.+b_.*x_))*(h_+i_.*Sin[d_.+e_.*x_])/(f_+g_.*Cos[d_.+e_.*x_]),x_Symbol] :=
  2*i*Int[F^(c*(a+b*x))*(Sin[d+e*x]/(f+g*Cos[d+e*x])),x] +
  Int[F^(c*(a+b*x))*((h-i*Sin[d+e*x])/(f+g*Cos[d+e*x])),x] /;
FreeQ[{F,a,b,c,d,e,f,g,h,i},x] && EqQ[f^2-g^2,0] && EqQ[h^2-i^2,0] && EqQ[g*h+f*i,0]
```

5: $\int F^{c\,u}\,\text{Trig}[v]^n\,dx$ when $u == a + b\,x \land v == d + e\,x$

Derivation: Algebraic normalization

Rule: If $u == a + b\,x \land v == d + e\,x$, then

$$\int F^{c\,u}\,\text{Trig}[v]^n\,dx \longrightarrow \int F^{c\,(a+b\,x)}\,\text{Trig}[d + e\,x]^n\,dx$$

Program code:

```
Int[F_^(c_.*u_)*G_[v_]^n_.,x_Symbol] :=
  Int[F^(c*ExpandToSum[u,x])*G[ExpandToSum[v,x]]^n,x] /;
FreeQ[{F,c,n},x] && TrigQ[G] && LinearQ[{u,v},x] && Not[LinearMatchQ[{u,v},x]]
```

6. $\int (f\,x)^m F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]^n\,dx$ when $n\in\mathbb{Z}^+$

1: $\int (f\,x)^m F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]^n\,dx$ when $n\in\mathbb{Z}^+\land m>0$

Derivation: Integration by parts

Note: Each term of the resulting integrand will be similar in form to the original integrand, but the degree of the monomial will be smaller by one.

Rule: If $n\in\mathbb{Z}^+\land m>0$, let $u=\int F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]^n\,dx$, then

$$\int (f\,x)^m F^{c\,(a+b\,x)}\,\text{Sin}[d+e\,x]^n\,dx \longrightarrow (f\,x)^m u - f\,m\int (f\,x)^{m-1} u\,dx$$

Program code:

```
Int[(f_.*x_)^m_.*F_^(c_.*(a_.+b_.*x_))*Sin[d_.+e_.*x_]^n_.,x_Symbol] :=
  Module[{u=IntHide[F^(c*(a+b*x))*Sin[d+e*x]^n,x]},
  Dist[(f*x)^m,u,x] - f*m*Int[(f*x)^(m-1)*u,x]] /;
FreeQ[{F,a,b,c,d,e,f},x] && IGtQ[n,0] && GtQ[m,0]
```

```
Int[(f_.*x_)^m_.*F_^(c_.*(a_.+b_.*x_))*Cos[d_.+e_.*x_]^n_.,x_Symbol] :=
  Module[{u=IntHide[F^(c*(a+b*x))*Cos[d+e*x]^n,x]},
  Dist[(f*x)^m,u,x] - f*m*Int[(f*x)^(m-1)*u,x]] /;
FreeQ[{F,a,b,c,d,e,f},x] && IGtQ[n,0] && GtQ[m,0]
```

2: $\int (f\,x)^m F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]\,dx$ when $m<-1$

Derivation: Integration by parts

Basis: $(f\,x)^m == \partial_x \frac{(f\,x)^{m+1}}{f\,(m+1)}$

Basis: $\partial_x\left(F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]\right) == e\,F^{c\,(a+b\,x)}\,\mathrm{Cos}[d+e\,x]+b\,c\,\mathrm{Log}[F]\,F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]$

Rule: If $m<-1$, then

$$\int (f\,x)^m F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]\,dx \longrightarrow$$
$$\frac{(f\,x)^{m+1}}{f\,(m+1)} F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]-\frac{e}{f\,(m+1)}\int (f\,x)^{m+1} F^{c\,(a+b\,x)}\,\mathrm{Cos}[d+e\,x]\,dx-\frac{b\,c\,\mathrm{Log}[F]}{f\,(m+1)}\int (f\,x)^{m+1} F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]\,dx$$

Program code:

```
Int[(f_.*x_)^m_*F_^(c_.*(a_.+b_.*x_))*Sin[d_.+e_.*x_],x_Symbol] :=
  (f*x)^(m+1)/(f*(m+1))*F^(c*(a+b*x))*Sin[d+e*x] -
  e/(f*(m+1))*Int[(f*x)^(m+1)*F^(c*(a+b*x))*Cos[d+e*x],x] -
  b*c*Log[F]/(f*(m+1))*Int[(f*x)^(m+1)*F^(c*(a+b*x))*Sin[d+e*x],x] /;
FreeQ[{F,a,b,c,d,e,f,m},x] && (LtQ[m,-1] || SumSimplerQ[m,1])
```

```
Int[(f_.*x_)^m_*F_^(c_.*(a_.+b_.*x_))*Cos[d_.+e_.*x_],x_Symbol] :=
  (f*x)^(m+1)/(f*(m+1))*F^(c*(a+b*x))*Cos[d+e*x] +
  e/(f*(m+1))*Int[(f*x)^(m+1)*F^(c*(a+b*x))*Sin[d+e*x],x] -
  b*c*Log[F]/(f*(m+1))*Int[(f*x)^(m+1)*F^(c*(a+b*x))*Cos[d+e*x],x] /;
FreeQ[{F,a,b,c,d,e,f,m},x] && (LtQ[m,-1] || SumSimplerQ[m,1])
```

x: $\int (f\,x)^m\,F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]^n\,dx$ when $n \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\mathrm{Sin}[z] == \frac{i}{2}\left(e^{-i\,z} - e^{i\,z}\right)$

Basis: $\mathrm{Cos}[z] == \frac{1}{2}\left(e^{-i\,z} + e^{i\,z}\right)$

Rule: If $n \in \mathbb{Z}^+$, then

$$\int (f\,x)^m\,F^{c\,(a+b\,x)}\,\mathrm{Sin}[d+e\,x]^n\,dx \longrightarrow \frac{i^n}{2^n}\int (f\,x)^m\,F^{c\,(a+b\,x)}\,\mathrm{ExpandIntegrand}\left[\left(e^{-i\,(d+e\,x)} - e^{i\,(d+e\,x)}\right)^n,\,x\right]dx$$

Program code:

```
(* Int[(f_.*x_)^m_.*F_^(c_.*(a_.+b_.*x_))*Sin[d_.+e_.*x_]^n_.,x_Symbol] :=
  I^n/2^n*Int[ExpandIntegrand[(f*x)^m*F^(c*(a+b*x)),(E^(-I*(d+e*x))-E^(I*(d+e*x)))^n,x],x] /;
FreeQ[{F,a,b,c,d,e,f},x] && IGtQ[n,0] *)
```

```
(* Int[(f_.*x_)^m_.*F_^(c_.*(a_.+b_.*x_))*Cos[d_.+e_.*x_]^n_.,x_Symbol] :=
  1/2^n*Int[ExpandIntegrand[(f*x)^m*F^(c*(a+b*x)),(E^(-I*(d+e*x))+E^(I*(d+e*x)))^n,x],x] /;
FreeQ[{F,a,b,c,d,e,f},x] && IGtQ[n,0] *)
```

7. $\int u\, F^{c\,(a+b\,x)}\, \text{Sin}[d+e\,x]^m\, \text{Cos}[f+g\,x]^n\, dx$

1: $\int F^{c\,(a+b\,x)}\, \text{Sin}[d+e\,x]^m\, \text{Cos}[f+g\,x]^n\, dx$ when $(m \mid n) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $(m \mid n) \in \mathbb{Z}^+$, then

$$\int F^{c\,(a+b\,x)}\, \text{Sin}[d+e\,x]^m\, \text{Cos}[f+g\,x]^n\, dx \longrightarrow \int F^{c\,(a+b\,x)}\, \text{TrigReduce}\left[\text{Sin}[d+e\,x]^m\, \text{Cos}[f+g\,x]^n\right] dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*Sin[d_.+e_.*x_]^m_.*Cos[f_.+g_.*x_]^n_.,x_Symbol] :=
  Int[ExpandTrigReduce[F^(c*(a+b*x)),Sin[d+e*x]^m*Cos[f+g*x]^n,x],x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && IGtQ[m,0] && IGtQ[n,0]
```

2: $\int x^p\, F^{c\,(a+b\,x)}\, \text{Sin}[d+e\,x]^m\, \text{Cos}[f+g\,x]^n\, dx$ when $(m \mid n \mid p) \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $(m \mid n \mid p) \in \mathbb{Z}^+$, then

$$\int x^p\, F^{c\,(a+b\,x)}\, \text{Sin}[d+e\,x]^m\, \text{Cos}[f+g\,x]^n\, dx \longrightarrow \int x^p\, F^{c\,(a+b\,x)}\, \text{TrigReduce}\left[\text{Sin}[d+e\,x]^m\, \text{Cos}[f+g\,x]^n\right] dx$$

Program code:

```
Int[x_^p_.*F_^(c_.*(a_.+b_.*x_))*Sin[d_.+e_.*x_]^m_.*Cos[f_.+g_.*x_]^n_.,x_Symbol] :=
  Int[ExpandTrigReduce[x^p*F^(c*(a+b*x)),Sin[d+e*x]^m*Cos[f+g*x]^n,x],x] /;
FreeQ[{F,a,b,c,d,e,f,g},x] && IGtQ[m,0] && IGtQ[n,0] && IGtQ[p,0]
```

8: $\int F^{c\,(a+b\,x)}\,\text{Trig}[d+e\,x]^m\,\text{Trig}[d+e\,x]^n\,dx$ when $(m\mid n)\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $(m\mid n)\in\mathbb{Z}^+$, then

$$\int F^{c\,(a+b\,x)}\,\text{Trig}[d+e\,x]^m\,\text{Trig}[d+e\,x]^n\,dx\;\longrightarrow\;\int F^{c\,(a+b\,x)}\,\text{TrigToExp}\left[\text{Trig}[d+e\,x]^m\,\text{Trig}[d+e\,x]^n,\,x\right]dx$$

Program code:

```
Int[F_^(c_.*(a_.+b_.*x_))*G_[d_.+e_.*x_]^m_.*H_[d_.+e_.*x_]^n_.,x_Symbol] :=
  Int[ExpandTrigToExp[F^(c*(a+b*x)),G[d+e*x]^m*H[d+e*x]^n,x],x] /;
FreeQ[{F,a,b,c,d,e},x] && IGtQ[m,0] && IGtQ[n,0] && TrigQ[G] && TrigQ[H]
```

9: $\int F^{a+b\,x+c\,x^2}\,\text{Sin}[d+e\,x+f\,x^2]^n\,dx$ when $n\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+$, then

$$\int F^{a+b\,x+c\,x^2}\,\text{Sin}[d+e\,x+f\,x^2]^n\,dx\;\longrightarrow\;\int F^{a+b\,x+c\,x^2}\,\text{TrigToExp}\left[\text{Sin}[d+e\,x+f\,x^2]^n\right]dx$$

Program code:

```
Int[F_^u_*Sin[v_]^n_.,x_Symbol] :=
  Int[ExpandTrigToExp[F^u,Sin[v]^n,x],x] /;
FreeQ[F,x] && (LinearQ[u,x] || PolyQ[u,x,2]) && (LinearQ[v,x] || PolyQ[v,x,2]) && IGtQ[n,0]
```

```
Int[F_^u_*Cos[v_]^n_.,x_Symbol] :=
  Int[ExpandTrigToExp[F^u,Cos[v]^n,x],x] /;
FreeQ[F,x] && (LinearQ[u,x] || PolyQ[u,x,2]) && (LinearQ[v,x] || PolyQ[v,x,2]) && IGtQ[n,0]
```

10: $\int F^{a+b\,x+c\,x^2}\,\text{Sin}\left[d+e\,x+f\,x^2\right]^m\,\text{Cos}\left[d+e\,x+f\,x^2\right]^n\,dx$ when $(m\mid n)\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $(m\mid n)\in\mathbb{Z}^+$, then

$$\int F^{a+b\,x+c\,x^2}\,\text{Sin}\left[d+e\,x+f\,x^2\right]^m\,\text{Cos}\left[d+e\,x+f\,x^2\right]^n\,dx\;\longrightarrow\;\int F^{a+b\,x+c\,x^2}\,\text{TrigToExp}\left[\text{Sin}\left[d+e\,x+f\,x^2\right]^m\,\text{Cos}\left[d+e\,x+f\,x^2\right]^n\right]\,dx$$

Program code:

```
Int[F_^u_*Sin[v_]^m_.*Cos[v_]^n_.,x_Symbol] :=
  Int[ExpandTrigToExp[F^u,Sin[v]^m*Cos[v]^n,x],x] /;
FreeQ[F,x] && (LinearQ[u,x] || PolyQ[u,x,2]) && (LinearQ[v,x] || PolyQ[v,x,2]) && IGtQ[m,0] && IGtQ[n,0]
```