

## Rules for integrands involving polylogarithms

$$1. \int u \operatorname{PolyLog}[n, a (b x^p)^q] dx$$

$$1. \int \operatorname{PolyLog}[n, a (b x^p)^q] dx$$

$$1. \int \operatorname{PolyLog}[n, a (b x^p)^q] dx \text{ when } n > 0$$

$$\textcolor{red}{x}: \int \operatorname{PolyLog}[2, a (b x^p)^q] dx$$

Derivation: Integration by parts

Note: This rule not necessary for host systems, like *Mathematica*, that automatically simplify  $\operatorname{PolyLog}[1, z]$  to  $-\operatorname{Log}[1 - z]$ .

Rule:

$$\int \operatorname{PolyLog}[2, a (b x^p)^q] dx \rightarrow x \operatorname{PolyLog}[2, a (b x^p)^q] + p q \int \operatorname{Log}[1 - a (b x^p)^q] dx$$

Program code:

```
(* Int[PolyLog[2,a.*(b_.**x_^p_.)^q_.],x_Symbol] :=
  x*PolyLog[2,a*(b*x^p)^q] + p*q*Int[Log[1-a*(b*x^p)^q],x] /;
FreeQ[{a,b,p,q},x] *)
```

$$\mathbf{2:} \int \text{PolyLog}[n, a (b x^p)^q] dx \text{ when } n > 0$$

Derivation: Integration by parts

Rule: If  $n > 0$ , then

$$\int \text{PolyLog}[n, a (b x^p)^q] dx \rightarrow x \text{PolyLog}[n, a (b x^p)^q] - p q \int \text{PolyLog}[n-1, a (b x^p)^q] dx$$

Program code:

```
Int[PolyLog[n_, a_.*(b_.*x_^p_.)^q_.], x_Symbol] :=
  x*PolyLog[n, a*(b*x^p)^q] - p*q*Int[PolyLog[n-1, a*(b*x^p)^q], x] /;
FreeQ[{a, b, p, q}, x] && GtQ[n, 0]
```

$$\mathbf{2:} \int \text{PolyLog}[n, a (b x^p)^q] dx \text{ when } n < -1$$

Derivation: Inverted integration by parts

Rule: If  $n < -1$ , then

$$\int \text{PolyLog}[n, a (b x^p)^q] dx \rightarrow \frac{x \text{PolyLog}[n+1, a (b x^p)^q]}{p q} - \frac{1}{p q} \int \text{PolyLog}[n+1, a (b x^p)^q] dx$$

Program code:

```
Int[PolyLog[n_, a_.*(b_.*x_^p_.)^q_.], x_Symbol] :=
  x*PolyLog[n+1, a*(b*x^p)^q]/(p*q) - 1/(p*q)*Int[PolyLog[n+1, a*(b*x^p)^q], x] /;
FreeQ[{a, b, p, q}, x] && LtQ[n, -1]
```

**U:**  $\int \text{PolyLog}[n, a (b x^p)^q] dx$

Rule:

$$\int \text{PolyLog}[n, a (b x^p)^q] dx \rightarrow \int \text{PolyLog}[n, a (b x^p)^q] dx$$

Program code:

```
Int[PolyLog[n_, a_.*(b_.*x_^p_)^q_], x_Symbol] :=
  Unintegrable[PolyLog[n, a*(b*x^p)^q], x] /;
  FreeQ[{a, b, n, p, q}, x]
```

2.  $\int (d x)^m \text{PolyLog}[n, a (b x^p)^q] dx$

**1:**  $\int \frac{\text{PolyLog}[n, a (b x^p)^q]}{x} dx$

Derivation: Primitive rule

Basis:  $\frac{\partial \text{Li}_n(z)}{\partial z} = \frac{\text{Li}_{n-1}(z)}{z}$

Rule:

$$\int \frac{\text{PolyLog}[n, a (b x^p)^q]}{x} dx \rightarrow \frac{\text{PolyLog}[n+1, a (b x^p)^q]}{p q}$$

Program code:

```
Int[PolyLog[n_, c_.*(a_.+b_.*x_)^p_]/(d_.*e_.*x_), x_Symbol] :=
  PolyLog[n+1, c*(a+b*x)^p]/(e*p) /;
  FreeQ[{a, b, c, d, e, n, p}, x] && EqQ[b*d, a*e]
```

```
Int[PolyLog[n_, a_.*(b_.*x_^p_.)^q_.]/x_, x_Symbol] :=
  PolyLog[n+1, a*(b*x^p)^q]/(p*q) /;
FreeQ[{a, b, n, p, q}, x]
```

2.  $\int (dx)^m \text{PolyLog}[n, a (bx^p)^q] dx$  when  $m \neq -1$

1:  $\int (dx)^m \text{PolyLog}[n, a (bx^p)^q] dx$  when  $m \neq -1 \wedge n > 0$

Derivation: Integration by parts

Rule: If  $m \neq -1 \wedge n > 0$ , then

$$\int (dx)^m \text{PolyLog}[n, a (bx^p)^q] dx \rightarrow \frac{(dx)^{m+1} \text{PolyLog}[n, a (bx^p)^q]}{d(m+1)} - \frac{pq}{m+1} \int (dx)^m \text{PolyLog}[n-1, a (bx^p)^q] dx$$

Program code:

```
Int[(d_.*x_)^m_.*PolyLog[n_, a_.*(b_.*x_^p_.)^q_.], x_Symbol] :=
  (d*x)^(m+1)*PolyLog[n, a*(b*x^p)^q]/(d*(m+1)) -
  p*q/(m+1)*Int[(d*x)^m*PolyLog[n-1, a*(b*x^p)^q], x] /;
FreeQ[{a, b, d, m, p, q}, x] && NeQ[m, -1] && GtQ[n, 0]
```

**2:**  $\int (d x)^m \text{PolyLog}[n, a (b x^p)^q] dx$  when  $m \neq -1 \wedge n < -1$

Derivation: Inverted integration by parts

Rule: If  $m \neq -1 \wedge n < -1$ , then

$$\int (d x)^m \text{PolyLog}[n, a (b x^p)^q] dx \rightarrow \frac{(d x)^{m+1} \text{PolyLog}[n+1, a (b x^p)^q]}{d p q} - \frac{m+1}{p q} \int (d x)^m \text{PolyLog}[n+1, a (b x^p)^q] dx$$

Program code:

```
Int[(d_.**x_)^m_.*PolyLog[n_,a_.*(b_.**x_^p_)^q_.],x_Symbol] :=
  (d*x)^(m+1)*PolyLog[n+1,a*(b*x^p)^q]/(d*p*q) -
  (m+1)/(p*q)*Int[(d*x)^m*PolyLog[n+1,a*(b*x^p)^q],x] /;
FreeQ[{a,b,d,m,p,q},x] && NeQ[m,-1] && LtQ[n,-1]
```

**U:**  $\int (d x)^m \text{PolyLog}[n, a (b x^p)^q] dx$

Rule:

$$\int (d x)^m \text{PolyLog}[n, a (b x^p)^q] dx \rightarrow \int (d x)^m \text{PolyLog}[n, a (b x^p)^q] dx$$

Program code:

```
Int[(d_.**x_)^m_.*PolyLog[n_,a_.*(b_.**x_^p_)^q_.],x_Symbol] :=
  Unintegrable[(d*x)^m*PolyLog[n,a*(b*x^p)^q],x] /;
FreeQ[{a,b,d,m,n,p,q},x]
```

$$3: \int \frac{\text{Log}[c x^m]^r \text{PolyLog}[n, a (b x^p)^q]}{x} dx \text{ when } r > 0$$

Derivation: Integration by parts

Rule: If  $r > 0$ , then

$$\int \frac{\text{Log}[c x^m]^r \text{PolyLog}[n, a (b x^p)^q]}{x} dx \rightarrow \frac{\text{Log}[c x^m]^r \text{PolyLog}[n+1, a (b x^p)^q]}{p q} - \frac{m r}{p q} \int \frac{\text{Log}[c x^m]^{r-1} \text{PolyLog}[n+1, a (b x^p)^q]}{x} dx$$

Program code:

```
Int[Log[c_.**x_^m_.]^r_.*PolyLog[n_,a_.*(b_.**x_^p_.)^q_.]/x_,x_Symbol] :=
  Log[c*x^m]^r*PolyLog[n+1,a*(b*x^p)^q]/(p*q) -
  m*r/(p*q)*Int[Log[c*x^m]^(r-1)*PolyLog[n+1,a*(b*x^p)^q]/x,x] /;
FreeQ[{a,b,c,m,n,q,r},x] && GtQ[r,0]
```

$$2. \int \text{PolyLog}[n, c (a + b x)^p] dx$$

$$1: \int \text{PolyLog}[n, c (a + b x)^p] dx \text{ when } n > 0$$

Derivation: Integration by parts and algebraic expansion

$$\text{Basis: } \partial_x \text{PolyLog}[n, c (a + b x)^p] == \frac{b p \text{PolyLog}[n-1, c (a + b x)^p]}{a + b x}$$

$$\text{Basis: } \frac{x}{a + b x} == \frac{1}{b} - \frac{a}{b (a + b x)}$$

Rule: If  $n > 0$ , then

$$\int \text{PolyLog}[n, c (a + b x)^p] dx \rightarrow x \text{PolyLog}[n, c (a + b x)^p] - b p \int \frac{x \text{PolyLog}[n-1, c (a + b x)^p]}{a + b x} dx \rightarrow$$

$$x \operatorname{PolyLog}[n, c (a + b x)^p] - p \int \operatorname{PolyLog}[n-1, c (a + b x)^p] dx + a p \int \frac{\operatorname{PolyLog}[n-1, c (a + b x)^p]}{a + b x} dx$$

### Program code:

```
Int[PolyLog[n_, c_.*(a_.+b_.*x_)^p_.], x_Symbol] :=
  x*PolyLog[n, c*(a+b*x)^p] -
  p*Int[PolyLog[n-1, c*(a+b*x)^p], x] +
  a*p*Int[PolyLog[n-1, c*(a+b*x)^p]/(a+b*x), x] /;
FreeQ[{a,b,c,p}, x] && GtQ[n, 0]
```

$$2. \int (d + e x)^m \text{PolyLog}[n, c (a + b x)^p] dx$$

$$1. \int (d + e x)^m \text{PolyLog}[2, c (a + b x)] dx$$

$$1. \int \frac{\text{PolyLog}[2, c (a + b x)]}{d + e x} dx$$

$$1: \int \frac{\text{PolyLog}[2, c (a + b x)]}{d + e x} dx \text{ when } c (b d - a e) + e = 0$$

Derivation: Integration by parts

Basis: If  $c (b d - a e) + e = 0$ , then  $\frac{1}{d + e x} = \partial_x \frac{\text{Log}[1 - a c - b c x]}{e}$

Basis:  $\partial_x \text{PolyLog}[2, c (a + b x)] = - \frac{b \text{Log}[1 - a c - b c x]}{a + b x}$

Rule: If  $c (b d - a e) + e = 0$ , then

$$\int \frac{\text{PolyLog}[2, c (a + b x)]}{d + e x} dx \rightarrow \frac{\text{Log}[1 - a c - b c x] \text{PolyLog}[2, c (a + b x)]}{e} + \frac{b}{e} \int \frac{\text{Log}[1 - a c - b c x]^2}{a + b x} dx$$

Program code:

```
Int[PolyLog[2,c.*(a_.+b_.*x_)]/(d_.+e_.*x_),x_Symbol] :=
  Log[1-a*c-b*c*x]*PolyLog[2,c*(a+b*x)]/e + b/e*Int[Log[1-a*c-b*c*x]^2/(a+b*x),x] /;
FreeQ[{a,b,c,d,e},x] && EqQ[c*(b*d-a*e)+e,0]
```



$$\text{2: } \int \frac{\text{PolyLog}[2, c(a+bx)]}{d+ex} dx \text{ when } c(bd-ae) + e \neq 0$$

Derivation: Integration by parts

$$\text{Basis: } \partial_x \text{PolyLog}[2, c(a+bx)] = -\frac{b \text{Log}[1-ac-bcx]}{a+bx}$$

Rule: If  $c(bd-ae) + e \neq 0$ , then

$$\int \frac{\text{PolyLog}[2, c(a+bx)]}{d+ex} dx \rightarrow \frac{\text{Log}[d+ex] \text{PolyLog}[2, c(a+bx)]}{e} + \frac{b}{e} \int \frac{\text{Log}[d+ex] \text{Log}[1-ac-bcx]}{a+bx} dx$$

Program code:

```
Int[PolyLog[2,c.*(a_.+b_.x_)]/(d_.+e_.x_),x_Symbol] :=
  Log[d+e*x]*PolyLog[2,c*(a+b*x)]/e + b/e*Int[Log[d+e*x]*Log[1-a*c-b*c*x]/(a+b*x),x] /;
FreeQ[{a,b,c,d,e},x] && NeQ[c*(b*d-a*e)+e,0]
```

$$\text{2: } \int (d+ex)^m \text{PolyLog}[2, c(a+bx)] dx \text{ when } m \neq -1$$

Derivation: Integration by parts

Rule: If  $m \neq -1$ , then

$$\int (d+ex)^m \text{PolyLog}[2, c(a+bx)] dx \rightarrow \frac{(d+ex)^{m+1} \text{PolyLog}[2, c(a+bx)]}{e(m+1)} + \frac{b}{e(m+1)} \int \frac{(d+ex)^{m+1} \text{Log}[1-ac-bcx]}{a+bx} dx$$

Program code:

```
Int[(d_.+e_.x_)^m_.*PolyLog[2,c.*(a_.+b_.x_)],x_Symbol] :=
  (d+e*x)^(m+1)*PolyLog[2,c*(a+b*x)]/(e*(m+1)) + b/(e*(m+1))*Int[(d+e*x)^(m+1)*Log[1-a*c-b*c*x]/(a+b*x),x] /;
FreeQ[{a,b,c,d,e,m},x] && NeQ[m,-1]
```

**x:**  $\int (d + e x)^m \text{PolyLog}[n, c (a + b x)^p] dx$  when  $n > 0 \wedge m \in \mathbb{Z}^+$

Derivation: Integration by parts

Rule: If  $n > 0 \wedge m \in \mathbb{Z}^+$ , then

$$\int (d + e x)^m \text{PolyLog}[n, c (a + b x)^p] dx \rightarrow \frac{(d + e x)^{m+1} \text{PolyLog}[n, c (a + b x)^p]}{e (m + 1)} - \frac{b p}{e (m + 1)} \int \frac{(d + e x)^{m+1} \text{PolyLog}[n - 1, c (a + b x)^p]}{a + b x} dx$$

Program code:

```
(* Int[(d_+e_*x_)^m_*PolyLog[n_,c_*(a_+b_*x_)^p_],x_Symbol] :=
  (d+e*x)^(m+1)*PolyLog[n,c*(a+b*x)^p]/(e*(m+1)) -
  b*p/(e*(m+1))*Int[(d+e*x)^(m+1)*PolyLog[n-1,c*(a+b*x)^p]/(a+b*x),x] /;
FreeQ[{a,b,c,d,e,m,p},x] && GtQ[n,0] && IGtQ[m,0] *)
```

$$\mathbf{2:} \int x^m \text{PolyLog}[n, c (a + b x)^p] dx \text{ when } n > 0 \wedge m \in \mathbb{Z} \wedge m \neq -1$$

Derivation: Integration by parts

Rule: If  $n > 0 \wedge m \in \mathbb{Z} \wedge m \neq -1$ , then

$$\int x^m \text{PolyLog}[n, c (a + b x)^p] dx \rightarrow -\frac{(a^{m+1} - b^{m+1} x^{m+1}) \text{PolyLog}[n, c (a + b x)^p]}{(m+1) b^{m+1}} + \frac{p}{(m+1) b^m} \int \text{PolyLog}[n-1, c (a + b x)^p] \text{ExpandIntegrand}\left[\frac{a^{m+1} - b^{m+1} x^{m+1}}{a + b x}, x\right] dx$$

Program code:

```
Int[x_^m_.*PolyLog[n_,c_.*(a_.+b_.*x_)^p_.],x_Symbol] :=
  -(a^(m+1)-b^(m+1)*x^(m+1))*PolyLog[n,c*(a+b*x)^p]/((m+1)*b^(m+1)) +
  p/((m+1)*b^m)*Int[ExpandIntegrand[PolyLog[n-1,c*(a+b*x)^p],(a^(m+1)-b^(m+1)*x^(m+1))/(a+b*x),x],x] /;
FreeQ[{a,b,c,p},x] && GtQ[n,0] && IntegerQ[m] && NeQ[m,-1]
```

$$3. \int u (g + h \text{Log}[f (d + e x)^n]) \text{PolyLog}[2, c (a + b x)] dx$$

$$\mathbf{1:} \int (g + h \text{Log}[f (d + e x)^n]) \text{PolyLog}[2, c (a + b x)] dx$$

Derivation: Integration by parts and algebraic expansion

$$\text{Basis: } \partial_x ((g + h \text{Log}[f (d + e x)^n]) \text{PolyLog}[2, c (a + b x)]) = -\frac{b (g + h \text{Log}[f (d + e x)^n]) \text{Log}[1 - c (a + b x)]}{a + b x} + \frac{e h n \text{PolyLog}[2, c (a + b x)]}{d + e x}$$

Rule:

$$\int (g + h \text{Log}[f (d + e x)^n]) \text{PolyLog}[2, c (a + b x)] dx \rightarrow x (g + h \text{Log}[f (d + e x)^n]) \text{PolyLog}[2, c (a + b x)] +$$

$$b \int (g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{Log}[1 - ac - bcx] \operatorname{ExpandIntegrand}\left[\frac{x}{a + bx}, x\right] dx - ehn \int \operatorname{PolyLog}[2, c(a + bx)] \operatorname{ExpandIntegrand}\left[\frac{x}{d + ex}, x\right] dx$$

Program code:

```
Int[(g_.+h_.*Log[f_.*(d_.+e_.*x_)^n_.])*PolyLog[2,c_.*(a_.+b_.*x_)],x_Symbol] :=
  x*(g+h*Log[f*(d+e*x)^n])*PolyLog[2,c*(a+b*x)] +
  b*Int[(g+h*Log[f*(d+e*x)^n])*Log[1-a*c-b*c*x]*ExpandIntegrand[x/(a+b*x),x],x] -
  e*h*n*Int[PolyLog[2,c*(a+b*x)]*ExpandIntegrand[x/(d+e*x),x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,n},x]
```

$$2. \int x^m (g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{PolyLog}[2, c(a + bx)] dx \text{ when } m \in \mathbb{Z}$$

$$1. \int \frac{(g + h \operatorname{Log}[1 + ex]) \operatorname{PolyLog}[2, cx]}{x} dx \text{ when } c + e = 0$$

$$1: \int \frac{\operatorname{Log}[1 + ex] \operatorname{PolyLog}[2, cx]}{x} dx \text{ when } c + e = 0$$

Derivation: Integration by substitution

Basis: If  $e + c = 0$ , then  $\frac{\operatorname{Log}[1+ex]}{x} = -\partial_x \operatorname{PolyLog}[2, cx]$

Rule: If  $c + e = 0$ , then

$$\int \frac{\operatorname{Log}[1 + ex] \operatorname{PolyLog}[2, cx]}{x} dx \rightarrow -\frac{\operatorname{PolyLog}[2, cx]^2}{2}$$

Program code:

```
Int[Log[1+e_.*x_]*PolyLog[2,c_.*x_]/x_,x_Symbol] :=
  -PolyLog[2,c*x]^2/2 /;
FreeQ[{c,e},x] && EqQ[c+e,0]
```

**2:**  $\int \frac{(g + h \operatorname{Log}[1 + e x]) \operatorname{PolyLog}[2, c x]}{x} dx$  when  $c + e == 0$

Derivation: Algebraic expansion

Rule: If  $c + e == 0$ , then

$$\int \frac{(g + h \operatorname{Log}[1 + e x]) \operatorname{PolyLog}[2, c x]}{x} dx \rightarrow g \int \frac{\operatorname{PolyLog}[2, c x]}{x} dx + h \int \frac{\operatorname{Log}[1 + e x] \operatorname{PolyLog}[2, c x]}{x} dx$$

Program code:

```
Int[(g_+h_.*Log[1+e_.*x_])*PolyLog[2,c_.*x_]/x_,x_Symbol] :=
  g*Int[PolyLog[2,c*x]/x,x] + h*Int[(Log[1+e*x])*PolyLog[2,c*x]/x,x] /;
FreeQ[{c,e,g,h},x] && EqQ[c+e,0]
```

$$\mathbf{2:} \int x^m (g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{PolyLog}[2, c(a + bx)] dx \text{ when } m \in \mathbb{Z} \wedge m \neq -1$$

Derivation: Integration by parts and algebraic expansion

$$\begin{aligned} \text{Basis: } \partial_x ((g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{PolyLog}[2, c(a + bx)]) = \\ - \frac{b(g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{Log}[1 - ac - bcx]}{a + bx} + \frac{e h n \operatorname{PolyLog}[2, c(a + bx)]}{d + ex} \end{aligned}$$

Rule: If  $m \in \mathbb{Z} \wedge m \neq -1$ , then

$$\begin{aligned} \int x^m (g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{PolyLog}[2, c(a + bx)] dx \rightarrow \\ \frac{x^{m+1} (g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{PolyLog}[2, c(a + bx)]}{m+1} + \\ \frac{b}{m+1} \int (g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{Log}[1 - ac - bcx] \operatorname{ExpandIntegrand}\left[\frac{x^{m+1}}{a + bx}, x\right] dx - \frac{e h n}{m+1} \int \operatorname{PolyLog}[2, c(a + bx)] \operatorname{ExpandIntegrand}\left[\frac{x^{m+1}}{d + ex}, x\right] dx \end{aligned}$$

Program code:

```
Int[x_^m.*(g_.+h_.*Log[f_.*(d_.+e_.*x_)^n_.])*PolyLog[2,c_.*(a_.+b_.*x_)],x_Symbol] :=
  x^(m+1)*(g+h*Log[f*(d+e*x)^n])*PolyLog[2,c*(a+b*x)]/(m+1) +
  b/(m+1)*Int[ExpandIntegrand[(g+h*Log[f*(d+e*x)^n])*Log[1-a*c-b*c*x],x^(m+1)/(a+b*x),x],x] -
  e*h*n/(m+1)*Int[ExpandIntegrand[PolyLog[2,c*(a+b*x)],x^(m+1)/(d+e*x),x],x] /;
FreeQ[{a,b,c,d,e,f,g,h,n},x] && IntegerQ[m] && NeQ[m,-1]
```

$$3: \int P[x] (g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{PolyLog}[2, c(a + bx)] dx$$

Derivation: Integration by parts and algebraic expansion

$$\text{Basis: } \partial_x ((g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{PolyLog}[2, c(a + bx)]) = \\ - \frac{b(g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{Log}[1 - ac - bcx]}{a + bx} + \frac{eh n \operatorname{PolyLog}[2, c(a + bx)]}{d + ex}$$

Rule: Let  $u \rightarrow \int P[x] dx$ , then

$$\int P[x] (g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{PolyLog}[2, c(a + bx)] dx \rightarrow \\ u(g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{PolyLog}[2, c(a + bx)] + \\ b \int (g + h \operatorname{Log}[f(d + ex)^n]) \operatorname{Log}[1 - ac - bcx] \operatorname{ExpandIntegrand}\left[\frac{u}{a + bx}, x\right] dx - eh n \int \operatorname{PolyLog}[2, c(a + bx)] \operatorname{ExpandIntegrand}\left[\frac{u}{d + ex}, x\right] dx$$

Program code:

```
Int[Px*(g_.+h_.*Log[f_.*(d_.+e_.*x_)^n_.])*PolyLog[2,c_.*(a_.+b_.*x_)],x_Symbol] :=
  With[{u=IntHide[Px,x]},
    u*(g+h*Log[f*(d+e*x)^n])*PolyLog[2,c*(a+b*x)] +
    b*Int[ExpandIntegrand[(g+h*Log[f*(d+e*x)^n])*Log[1-a*c-b*c*x],u/(a+b*x),x],x] -
    e*h*n*Int[ExpandIntegrand[PolyLog[2,c*(a+b*x)],u/(d+e*x),x],x] /;
  FreeQ[{a,b,c,d,e,f,g,h,n},x] && PolyQ[Px,x]
```

$$4. \int x^m P[x] (g + h \log[f(d + ex)^n]) \text{PolyLog}[2, c(a + bx)] dx \text{ when } m \in \mathbb{Z}$$

$$1: \int x^m P[x] (g + h \log[1 + ex]) \text{PolyLog}[2, cx] dx \text{ when } m \in \mathbb{Z}^- \wedge c + e == 0 \wedge P[x, -m - 1] \neq 0$$

Derivation: Algebraic expansion

Note: Separates out the term in the integrand of the form  $\frac{P[x, -m - 1] (g + h \log[1 + ex]) \text{PolyLog}[2, cx]}{x}$ .

Rule: If  $m \in \mathbb{Z}^- \wedge c + e == 0 \wedge P[x, -m - 1] \neq 0$ , then

$$\int x^m P[x] (g + h \log[1 + ex]) \text{PolyLog}[2, cx] dx \rightarrow \\ P[x, -m - 1] \int \frac{(g + h \log[1 + ex]) \text{PolyLog}[2, cx]}{x} dx + \int x^m (P[x] - P[x, -m - 1] x^{-m-1}) (g + h \log[1 + ex]) \text{PolyLog}[2, cx] dx$$

Program code:

```
Int[x^m_*Px_*(g_+h_*Log[1+e_*x_])*PolyLog[2,c_*x_],x_Symbol] :=
  Coeff[Px,x,-m-1]*Int[(g+h*Log[1+e*x])*PolyLog[2,c*x]/x,x] +
  Int[x^m*(Px-Coeff[Px,x,-m-1]*x^(-m-1))*(g+h*Log[1+e*x])*PolyLog[2,c*x],x] /;
FreeQ[{c,e,g,h},x] && PolyQ[Px,x] && ILtQ[m,0] && EqQ[c+e,0] && NeQ[Coeff[Px,x,-m-1],0]
```

$$2: \int x^m P[x] (g + h \log[f(d + ex)^n]) \text{PolyLog}[2, c(a + bx)] dx \text{ when } m \in \mathbb{Z}$$

Derivation: Integration by parts and algebraic expansion

$$\text{Basis: } \partial_x ((g + h \log[f(d + ex)^n]) \text{PolyLog}[2, c(a + bx)]) = \\ - \frac{b (g + h \log[f(d + ex)^n]) \log[1 - ac - bcx]}{a + bx} + \frac{e h n \text{PolyLog}[2, c(a + bx)]}{d + ex}$$

Rule: If  $m \in \mathbb{Z}$ , let  $u \rightarrow \int x^m P[x] dx$ , then

$$\int x^m P[x] (g + h \log[f(d + ex)^n]) \text{PolyLog}[2, c(a + bx)] dx \rightarrow$$



$$b \int (g + h \log[f(d + ex)^n]) \log[1 - ac - bcx] \operatorname{ExpandIntegrand}\left[\frac{u}{a + bx}, x\right] dx - ehn \int \operatorname{PolyLog}[2, c(a + bx)] \operatorname{ExpandIntegrand}\left[\frac{u}{d + ex}, x\right] dx$$

Program code:

```
Int[x^m_*Px_*(g_+h_*Log[f_*(d_+e_*x_)^n_])*PolyLog[2,c_*(a_+b_*x_)],x_Symbol] :=
  With[{u=IntHide[x^m*Px,x]},
    u*(g+h*Log[f*(d+e*x)^n])*PolyLog[2,c*(a+b*x)] +
    b*Int[ExpandIntegrand[(g+h*Log[f*(d+e*x)^n])*Log[1-a*c-b*c*x],u/(a+b*x),x],x] -
    e*h*n*Int[ExpandIntegrand[PolyLog[2,c*(a+b*x)],u/(d+e*x),x],x] /;
  FreeQ[{a,b,c,d,e,f,g,h,n},x] && PolyQ[Px,x] && IntegerQ[m]
```

**U:**  $\int x^m P[x] (g + h \log[f(d + ex)^n]) \operatorname{PolyLog}[2, c(a + bx)] dx$

Rule:

$$\int x^m P[x] (g + h \log[f(d + ex)^n]) \operatorname{PolyLog}[2, c(a + bx)] dx \rightarrow \int x^m P[x] (g + h \log[f(d + ex)^n]) \operatorname{PolyLog}[2, c(a + bx)] dx$$

Program code:

```
Int[x^m_*Px_*(g_+h_*Log[f_*(d_+e_*x_)^n_])*PolyLog[2,c_*(a_+b_*x_)],x_Symbol] :=
  Unintegrable[x^m*Px*(g+h*Log[f*(d+e*x)^n])*PolyLog[2,c*(a+b*x)],x] /;
  FreeQ[{a,b,c,d,e,f,g,h,m,n},x] && PolyQ[Px,x]
```

$$4. \int u \operatorname{PolyLog}[n, d (F^{c(a+bx)})^p] dx$$

$$1: \int \operatorname{PolyLog}[n, d (F^{c(a+bx)})^p] dx$$

Derivation: Primitive rule

$$\text{Basis: } \partial_z \operatorname{PolyLog}[n, z] = \frac{\operatorname{PolyLog}[n-1, z]}{z}$$

Rule:

$$\int \operatorname{PolyLog}[n, d (F^{c(a+bx)})^p] dx \rightarrow \frac{\operatorname{PolyLog}[n+1, d (F^{c(a+bx)})^p]}{b c p \operatorname{Log}[F]}$$

Program code:

```
Int[PolyLog[n_, d_.*(F_^(c_.*(a_.*b_.*x_)))^p_], x_Symbol] :=
  PolyLog[n+1, d*(F^(c*(a+b*x)))^p]/(b*c*p*Log[F]) /;
FreeQ[{F, a, b, c, d, n, p}, x]
```

**2:**  $\int (e + f x)^m \text{PolyLog}[n, d (F^c (a + b x))^p] dx$  when  $m > 0$

Derivation: Integration by parts

■ Basis:  $\text{PolyLog}[n, d (F^c (a + b x))^p] = \partial_x \frac{\text{PolyLog}[n+1, d (F^c (a + b x))^p]}{b c p \text{Log}[F]}$

Rule: If  $m > 0$ , then

$$\int (e + f x)^m \text{PolyLog}[n, d (F^c (a + b x))^p] dx \rightarrow \frac{(e + f x)^m \text{PolyLog}[n+1, d (F^c (a + b x))^p]}{b c p \text{Log}[F]} - \frac{f m}{b c p \text{Log}[F]} \int (e + f x)^{m-1} \text{PolyLog}[n+1, d (F^c (a + b x))^p] dx$$

Program code:

```
Int[(e_+f_*x_)^m_*PolyLog[n_,d_*(F^(c_*(a_+b_*x_)))^p_],x_Symbol] :=
  (e+f*x)^m*PolyLog[n+1,d*(F^(c*(a+b*x)))^p]/(b*c*p*Log[F]) -
  f*m/(b*c*p*Log[F])*Int[(e+f*x)^(m-1)*PolyLog[n+1,d*(F^(c*(a+b*x)))^p],x] /;
FreeQ[{F,a,b,c,d,e,f,n,p},x] && GtQ[m,0]
```

$$5. \int u \frac{\text{PolyLog}[n, F[x]] F'[x]}{F[x]} dx$$

$$1: \int \frac{\text{PolyLog}[n, F[x]] F'[x]}{F[x]} dx$$

$$\text{Basis: } \partial_x \text{PolyLog}[n+1, x] == \frac{\text{PolyLog}[n, x]}{x}$$

Rule:

$$\int \frac{\text{PolyLog}[n, F[x]] F'[x]}{F[x]} dx \rightarrow \text{PolyLog}[n+1, F[x]]$$

Program code:

```
Int[u_*PolyLog[n_,v_],x_Symbol] :=
  With[{w=DerivativeDivides[v,u*v,x]},
    w*PolyLog[n+1,v] /;
    Not[FalseQ[w]] /;
    FreeQ[n,x]
```

$$2: \int \frac{\text{Log}[G[x]] \text{PolyLog}[n, F[x]] F'[x]}{F[x]} dx$$

Derivation: Integration by parts

$$\text{Basis: } \frac{\text{PolyLog}[n, x]}{x} = \partial_x \text{PolyLog}[n+1, x]$$

Rule:

$$\int \frac{\text{Log}[G[x]] \text{PolyLog}[n, F[x]] F'[x]}{F[x]} dx \rightarrow \text{Log}[G[x]] \text{PolyLog}[n+1, F[x]] - \int \frac{G'[x] \text{PolyLog}[n+1, F[x]]}{G[x]} dx$$

Program code:

```
Int[u_*Log[w]*PolyLog[n_,v_],x_Symbol] :=
  With[{z=DerivativeDivides[v,u*v,x]},
    z*Log[w]*PolyLog[n+1,v] -
    Int[SimplifyIntegrand[z*D[w,x]*PolyLog[n+1,v]/w,x],x] /;
    Not[FalseQ[z]] /;
    FreeQ[n,x] && InverseFunctionFreeQ[w,x]
```