Rules for integrands of the form $(c + d\,x)^m\,\text{Hyper}[a + b\,x]^n\,\text{Hyper}[a + b\,x]^p$

1. $\int (c + d\,x)^m\,\text{Hyper}[a + b\,x]^n\,\text{Hyper}[a + b\,x]^p\,dx$

   1. $\int (c + d\,x)^m\,\text{Sinh}[a + b\,x]^n\,\text{Cosh}[a + b\,x]^p\,dx$

      1: $\int (c + d\,x)^m\,\text{Sinh}[a + b\,x]^n\,\text{Cosh}[a + b\,x]\,dx$ when $m \in \mathbb{Z}^+ \land n \neq -1$

Derivation: Integration by parts

Basis: $\text{Sinh}[a + b\,x]^n\,\text{Cosh}[a + b\,x] == \partial_x \frac{\text{Sinh}[a+b\,x]^{n+1}}{b\,(n+1)}$

Rule: If $m \in \mathbb{Z}^+ \land n \neq -1$, then

$$\int (c + d\,x)^m\,\text{Sinh}[a + b\,x]^n\,\text{Cosh}[a + b\,x]\,dx \longrightarrow \frac{(c + d\,x)^m\,\text{Sinh}[a + b\,x]^{n+1}}{b\,(n + 1)} - \frac{d\,m}{b\,(n + 1)} \int (c + d\,x)^{m-1}\,\text{Sinh}[a + b\,x]^{n+1}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sinh[a_.+b_.*x_]^n_.*Cosh[a_.+b_.*x_],x_Symbol] :=
  (c+d*x)^m*Sinh[a+b*x]^(n+1)/(b*(n+1)) -
  d*m/(b*(n+1))*Int[(c+d*x)^(m-1)*Sinh[a+b*x]^(n+1),x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

```
Int[(c_.+d_.*x_)^m_.*Sinh[a_.+b_.*x_]*Cosh[a_.+b_.*x_]^n_.,x_Symbol] :=
  (c+d*x)^m*Cosh[a+b*x]^(n+1)/(b*(n+1)) -
  d*m/(b*(n+1))*Int[(c+d*x)^(m-1)*Cosh[a+b*x]^(n+1),x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

2: $\int (c+d\,x)^m \operatorname{Sinh}[a+b\,x]^n \operatorname{Cosh}[a+b\,x]^p\,dx$ when $n\in\mathbb{Z}^+ \land p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $n\in\mathbb{Z}^+ \land p\in\mathbb{Z}^+$, then

$$\int (c+d\,x)^m \operatorname{Sinh}[a+b\,x]^n \operatorname{Cosh}[a+b\,x]^p\,dx \longrightarrow \int (c+d\,x)^m \operatorname{TrigReduce}\left[\operatorname{Sinh}[a+b\,x]^n \operatorname{Cosh}[a+b\,x]^p\right]dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sinh[a_.+b_.*x_]^n_.*Cosh[a_.+b_.*x_]^p_.,x_Symbol] :=
  Int[ExpandTrigReduce[(c+d*x)^m,Sinh[a+b*x]^n*Cosh[a+b*x]^p,x],x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && IGtQ[p,0]
```

2: $\int (c+d\,x)^m \operatorname{Sinh}[a+b\,x]^n \operatorname{Tanh}[a+b\,x]^p\,dx$ when $n\in\mathbb{Z}^+ \land p\in\mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\operatorname{Sinh}[z]^2 \operatorname{Tanh}[z]^2 == \operatorname{Sinh}[z]^2 - \operatorname{Tanh}[z]^2$

Rule: If $n\in\mathbb{Z}^+ \land p\in\mathbb{Z}^+$, then

$$\int (c+d\,x)^m \operatorname{Sinh}[a+b\,x]^n \operatorname{Tanh}[a+b\,x]^p\,dx \longrightarrow$$
$$\int (c+d\,x)^m \operatorname{Sinh}[a+b\,x]^n \operatorname{Tanh}[a+b\,x]^{p-2}\,dx - \int (c+d\,x)^m \operatorname{Sinh}[a+b\,x]^{n-2} \operatorname{Tanh}[a+b\,x]^p\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sinh[a_.+b_.*x_]^n_.*Tanh[a_.+b_.*x_]^p_.,x_Symbol] :=
  Int[(c+d*x)^m*Sinh[a+b*x]^n*Tanh[a+b*x]^(p-2),x] - Int[(c+d*x)^m*Sinh[a+b*x]^(n-2)*Tanh[a+b*x]^p,x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && IGtQ[p,0]
```

```
Int[(c_.+d_.*x_)^m_.*Cosh[a_.+b_.*x_]^n_.*Coth[a_.+b_.*x_]^p_.,x_Symbol] :=
  Int[(c+d*x)^m*Cosh[a+b*x]^n*Coth[a+b*x]^(p-2),x] + Int[(c+d*x)^m*Cosh[a+b*x]^(n-2)*Coth[a+b*x]^p,x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[n,0] && IGtQ[p,0]
```

3. $\int (c+d\,x)^m\,\mathrm{Sech}[a+b\,x]^n\,\mathrm{Tanh}[a+b\,x]^p\,dx$

1: $\int (c+d\,x)^m\,\mathrm{Sech}[a+b\,x]^n\,\mathrm{Tanh}[a+b\,x]\,dx$ when $m>0$

Derivation: Integration by parts

Basis: $\mathrm{Sech}[a+b\,x]^n\,\mathrm{Tanh}[a+b\,x] = -\partial_x \frac{\mathrm{Sech}[a+b\,x]^n}{b\,n}$

Note: Dummy exponent $p === 1$ required in program code so InputForm of integrand is recognized.

Rule: If $m>0$, then

$$\int (c+d\,x)^m\,\mathrm{Sech}[a+b\,x]^n\,\mathrm{Tanh}[a+b\,x]\,dx \to -\frac{(c+d\,x)^m\,\mathrm{Sech}[a+b\,x]^n}{b\,n}+\frac{d\,m}{b\,n}\int (c+d\,x)^{m-1}\,\mathrm{Sech}[a+b\,x]^n\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sech[a_.+b_.*x_]^n_.*Tanh[a_.+b_.*x_]^p_.,x_Symbol] :=
  -(c+d*x)^m*Sech[a+b*x]^n/(b*n) +
  d*m/(b*n)*Int[(c+d*x)^(m-1)*Sech[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x] && EqQ[p,1] && GtQ[m,0]
```

```
Int[(c_.+d_.*x_)^m_.*Csch[a_.+b_.*x_]^n_.*Coth[a_.+b_.*x_]^p_.,x_Symbol] :=
  -(c+d*x)^m*Csch[a+b*x]^n/(b*n) +
  d*m/(b*n)*Int[(c+d*x)^(m-1)*Csch[a+b*x]^n,x] /;
FreeQ[{a,b,c,d,n},x] && EqQ[p,1] && GtQ[m,0]
```

2: $\int (c+d x)^m \operatorname{Sech}[a+b x]^2 \operatorname{Tanh}[a+b x]^n \, dx$ when $m \in \mathbb{Z}^+ \land n \neq -1$

Derivation: Integration by parts

Basis: $\operatorname{Sech}[a+b x]^2 \operatorname{Tanh}[a+b x]^n == \partial_x \frac{\operatorname{Tanh}[a+b x]^{n+1}}{b (n+1)}$

Rule: If $m \in \mathbb{Z}^+ \land n \neq -1$, then

$$\int (c+d x)^m \operatorname{Sech}[a+b x]^2 \operatorname{Tanh}[a+b x]^n \, dx \rightarrow \frac{(c+d x)^m \operatorname{Tanh}[a+b x]^{n+1}}{b (n+1)} - \frac{d m}{b (n+1)} \int (c+d x)^{m-1} \operatorname{Tanh}[a+b x]^{n+1} \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sech[a_.+b_.*x_]^2*Tanh[a_.+b_.*x_]^n_.,x_Symbol] :=
  (c+d*x)^m*Tanh[a+b*x]^(n+1)/(b*(n+1)) -
  d*m/(b*(n+1))*Int[(c+d*x)^(m-1)*Tanh[a+b*x]^(n+1),x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

```
Int[(c_.+d_.*x_)^m_.*Csch[a_.+b_.*x_]^2*Coth[a_.+b_.*x_]^n_.,x_Symbol] :=
  -(c+d*x)^m*Coth[a+b*x]^(n+1)/(b*(n+1)) +
  d*m/(b*(n+1))*Int[(c+d*x)^(m-1)*Coth[a+b*x]^(n+1),x] /;
FreeQ[{a,b,c,d,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

3: $\int (c+d x)^m \operatorname{Sech}[a+b x]^n \operatorname{Tanh}[a+b x]^p \, dx$ when $\frac{p}{2} \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Basis: $\operatorname{Tanh}[z]^2 == 1 - \operatorname{Sech}[z]^2$

Rule: If $\frac{p}{2} \in \mathbb{Z}^+$, then

$$\int (c+d x)^m \operatorname{Sech}[a+b x]^n \operatorname{Tanh}[a+b x]^p \, dx \rightarrow$$

$$\int (c + d x)^m \,\text{Sech}[a + b x]^n \,\text{Tanh}[a + b x]^{p-2} \,dx - \int (c + d x)^m \,\text{Sech}[a + b x]^{n+2} \,\text{Tanh}[a + b x]^{p-2} \,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sech[a_.+b_.*x_]*Tanh[a_.+b_.*x_]^p_,x_Symbol] :=
  Int[(c+d*x)^m*Sech[a+b*x]*Tanh[a+b*x]^(p-2),x] - Int[(c+d*x)^m*Sech[a+b*x]^3*Tanh[a+b*x]^(p-2),x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[p/2,0]
```

```
Int[(c_.+d_.*x_)^m_.*Sech[a_.+b_.*x_]^n_.*Tanh[a_.+b_.*x_]^p_,x_Symbol] :=
  Int[(c+d*x)^m*Sech[a+b*x]^n*Tanh[a+b*x]^(p-2),x] - Int[(c+d*x)^m*Sech[a+b*x]^(n+2)*Tanh[a+b*x]^(p-2),x] /;
FreeQ[{a,b,c,d,m,n},x] && IGtQ[p/2,0]
```

```
Int[(c_.+d_.*x_)^m_.*Csch[a_.+b_.*x_]*Coth[a_.+b_.*x_]^p_,x_Symbol] :=
  Int[(c+d*x)^m*Csch[a+b*x]*Coth[a+b*x]^(p-2),x] + Int[(c+d*x)^m*Csch[a+b*x]^3*Coth[a+b*x]^(p-2),x] /;
FreeQ[{a,b,c,d,m},x] && IGtQ[p/2,0]
```

```
Int[(c_.+d_.*x_)^m_.*Csch[a_.+b_.*x_]^n_.*Coth[a_.+b_.*x_]^p_,x_Symbol] :=
  Int[(c+d*x)^m*Csch[a+b*x]^n*Coth[a+b*x]^(p-2),x] + Int[(c+d*x)^m*Csch[a+b*x]^(n+2)*Coth[a+b*x]^(p-2),x] /;
FreeQ[{a,b,c,d,m,n},x] && IGtQ[p/2,0]
```

4: $\int (c+d\,x)^m\,\mathrm{Sech}[a+b\,x]^n\,\mathrm{Tanh}[a+b\,x]^p\,dx$ when $m\in\mathbb{Z}^+\wedge\left(\frac{n}{2}\in\mathbb{Z}\vee\frac{p+1}{2}\in\mathbb{Z}\right)$

Derivation: Integration by parts

Rule: If $m\in\mathbb{Z}^+\wedge\left(\frac{n}{2}\in\mathbb{Z}\vee\frac{p+1}{2}\in\mathbb{Z}\right)$, let $u=\int\mathrm{Sech}[a+b\,x]^n\,\mathrm{Tanh}[a+b\,x]^p\,dx$, then

$$\int (c+d\,x)^m\,\mathrm{Sech}[a+b\,x]^n\,\mathrm{Tanh}[a+b\,x]^p\,dx\,\longrightarrow\,u\,(c+d\,x)^m-d\,m\int u\,(c+d\,x)^{m-1}\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Sech[a_.+b_.*x_]^n_.*Tanh[a_.+b_.*x_]^p_.,x_Symbol] :=
  With[{u=IntHide[Sech[a+b*x]^n*Tanh[a+b*x]^p,x]},
  Dist[(c+d*x)^m,u,x] - d*m*Int[(c+d*x)^(m-1)*u,x]] /;
FreeQ[{a,b,c,d,n,p},x] && IGtQ[m,0] && (IntegerQ[n/2] || IntegerQ[(p-1)/2])
```

```
Int[(c_.+d_.*x_)^m_.*Csch[a_.+b_.*x_]^n_.*Coth[a_.+b_.*x_]^p_.,x_Symbol] :=
  With[{u=IntHide[Csch[a+b*x]^n*Coth[a+b*x]^p,x]},
  Dist[(c+d*x)^m,u,x] - d*m*Int[(c+d*x)^(m-1)*u,x]] /;
FreeQ[{a,b,c,d,n,p},x] && IGtQ[m,0] && (IntegerQ[n/2] || IntegerQ[(p-1)/2])
```

4. $\int (c+d x)^m \operatorname{Sech}[a+b x]^p \operatorname{Csch}[a+b x]^n \, dx$

1: $\int (c+d x)^m \operatorname{Csch}[a+b x]^n \operatorname{Sech}[a+b x]^n \, dx$ when $n \in \mathbb{Z}$

Derivation: Algebraic simplification

Basis: $\operatorname{Csch}[z] \operatorname{Sech}[z] == 2 \operatorname{Csch}[2 z]$

Rule: If $n \in \mathbb{Z}$, then

$$\int (c+d x)^m \operatorname{Csch}[a+b x]^n \operatorname{Sech}[a+b x]^n \, dx \longrightarrow 2^n \int (c+d x)^m \operatorname{Csch}[2 a+2 b x]^n \, dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Csch[a_.+b_.*x_]^n_.*Sech[a_.+b_.*x_]^n_., x_Symbol] :=
  2^n*Int[(c+d*x)^m*Csch[2*a+2*b*x]^n,x] /;
FreeQ[{a,b,c,d},x] && RationalQ[m] && IntegerQ[n]
```

2: $\int (c+d\,x)^m \operatorname{Csch}[a+b\,x]^n \operatorname{Sech}[a+b\,x]^p\,dx$ when $(n \mid p) \in \mathbb{Z} \wedge m>0 \wedge n \neq p$

Derivation: Integration by parts

Rule: If $(n \mid p) \in \mathbb{Z} \wedge m>0 \wedge n \neq p$, let $u = \int \operatorname{Csch}[a+b\,x]^n \operatorname{Sech}[a+b\,x]^p\,dx$, then

$$\int (c+d\,x)^m \operatorname{Csch}[a+b\,x]^n \operatorname{Sech}[a+b\,x]^p\,dx \rightarrow (c+d\,x)^m\,u - d\,m \int (c+d\,x)^{m-1}\,u\,dx$$

Program code:

```
Int[(c_.+d_.*x_)^m_.*Csch[a_.+b_.*x_]^n_.*Sech[a_.+b_.*x_]^p_., x_Symbol] :=
  With[{u=IntHide[Csch[a+b*x]^n*Sech[a+b*x]^p,x]},
  Dist[(c+d*x)^m,u,x] - d*m*Int[(c+d*x)^(m-1)*u,x]] /;
FreeQ[{a,b,c,d},x] && IntegersQ[n,p] && GtQ[m,0] && NeQ[n,p]
```

5: $\int u^m \operatorname{Hyper}[v]^n \operatorname{Hyper}[w]^p\,dx$ when $u == c+d\,x \wedge v == w == a+b\,x$

Derivation: Algebraic normalization

Rule: If $u == c+d\,x \wedge v == w == a+b\,x$, then

$$\int u^m \operatorname{Hyper}[v]^n \operatorname{Hyper}[w]^p\,dx \rightarrow \int (c+d\,x)^m \operatorname{Hyper}[a+b\,x]^n \operatorname{Hyper}[a+b\,x]^p\,dx$$

Program code:

```
Int[u_^m_.*F_[v_]^n_.*G_[w_]^p_.,x_Symbol] :=
  Int[ExpandToSum[u,x]^m*F[ExpandToSum[v,x]]^n*G[ExpandToSum[v,x]]^p,x] /;
FreeQ[{m,n,p},x] && HyperbolicQ[F] && HyperbolicQ[G] && EqQ[v,w] && LinearQ[{u,v,w},x] && Not[LinearMatchQ[{u,v,w},x]]
```

2: $\int (e+f\,x)^m \operatorname{Cosh}[c+d\,x]\,(a+b\operatorname{Sinh}[c+d\,x])^n\,dx$ when $m\in\mathbb{Z}^+\wedge n\neq -1$

Derivation: Integration by parts

Basis: $\operatorname{Cosh}[c+d\,x]\,(a+b\operatorname{Sinh}[c+d\,x])^n = \partial_x \frac{(a+b\operatorname{Sinh}[c+d\,x])^{n+1}}{b\,d\,(n+1)}$

Rule: If $m\in\mathbb{Z}^+\wedge n\neq -1$, then

$$\int (e+f\,x)^m \operatorname{Cosh}[c+d\,x]\,(a+b\operatorname{Sinh}[c+d\,x])^n\,dx \to \frac{(e+f\,x)^m\,(a+b\operatorname{Sinh}[c+d\,x])^{n+1}}{b\,d\,(n+1)} - \frac{f\,m}{b\,d\,(n+1)}\int (e+f\,x)^{m-1}\,(a+b\operatorname{Sinh}[c+d\,x])^{n+1}\,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Cosh[c_.+d_.*x_]*(a_+b_.*Sinh[c_.+d_.*x_])^n_.,x_Symbol] :=
  (e+f*x)^m*(a+b*Sinh[c+d*x])^(n+1)/(b*d*(n+1)) -
  f*m/(b*d*(n+1))*Int[(e+f*x)^(m-1)*(a+b*Sinh[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

```
Int[(e_.+f_.*x_)^m_.*Sinh[c_.+d_.*x_]*(a_+b_.*Cosh[c_.+d_.*x_])^n_.,x_Symbol] :=
  (e+f*x)^m*(a+b*Cosh[c+d*x])^(n+1)/(b*d*(n+1)) -
  f*m/(b*d*(n+1))*Int[(e+f*x)^(m-1)*(a+b*Cosh[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

3: $\int (e+f\,x)^m \operatorname{Sech}[c+d\,x]^2\,(a+b\operatorname{Tanh}[c+d\,x])^n\,dx$ when $m\in\mathbb{Z}^+\wedge n\neq -1$

Derivation: Integration by parts

Basis: $\operatorname{Sech}[c+d\,x]^2\,(a+b\operatorname{Tanh}[c+d\,x])^n = \partial_x \frac{(a+b\operatorname{Tanh}[c+d\,x])^{n+1}}{b\,d\,(n+1)}$

Rule: If $m\in\mathbb{Z}^+\wedge n\neq -1$, then

$$\int (e+f x)^m \operatorname{Sech}[c+d x]^2 (a+b \operatorname{Tanh}[c+d x])^n \, dx \longrightarrow \frac{(e+f x)^m (a+b \operatorname{Tanh}[c+d x])^{n+1}}{b d (n+1)} - \frac{f m}{b d (n+1)} \int (e+f x)^{m-1} (a+b \operatorname{Tanh}[c+d x])^{n+1} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sech[c_.+d_.*x_]^2*(a_+b_.*Tanh[c_.+d_.*x_])^n_.,x_Symbol] :=
  (e+f*x)^m*(a+b*Tanh[c+d*x])^(n+1)/(b*d*(n+1)) -
  f*m/(b*d*(n+1))*Int[(e+f*x)^(m-1)*(a+b*Tanh[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

```
Int[(e_.+f_.*x_)^m_.*Csch[c_.+d_.*x_]^2*(a_+b_.*Coth[c_.+d_.*x_])^n_.,x_Symbol] :=
  -(e+f*x)^m*(a+b*Coth[c+d*x])^(n+1)/(b*d*(n+1)) +
  f*m/(b*d*(n+1))*Int[(e+f*x)^(m-1)*(a+b*Coth[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

4: $\int (e+f x)^m \operatorname{Sech}[c+d x] \operatorname{Tanh}[c+d x] (a+b \operatorname{Sech}[c+d x])^n \, dx$ when $m \in \mathbb{Z}^+ \land n \neq -1$

Derivation: Integration by parts

Basis: $\operatorname{Sech}[c+d x] \operatorname{Tanh}[c+d x] (a+b \operatorname{Sech}[c+d x])^n = -\partial_x \frac{(a+b \operatorname{Sech}[c+d x])^{n+1}}{b d (n+1)}$

Rule: If $m \in \mathbb{Z}^+ \land n \neq -1$, then

$$\int (e+f x)^m \operatorname{Sech}[c+d x] \operatorname{Tanh}[c+d x] (a+b \operatorname{Sech}[c+d x])^n \, dx \longrightarrow -\frac{(e+f x)^m (a+b \operatorname{Sech}[c+d x])^{n+1}}{b d (n+1)} + \frac{f m}{b d (n+1)} \int (e+f x)^{m-1} (a+b \operatorname{Sech}[c+d x])^{n+1} \, dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sech[c_.+d_.*x_]*Tanh[c_.+d_.*x_]*(a_+b_.*Sech[c_.+d_.*x_])^n_.,x_Symbol] :=
  -(e+f*x)^m*(a+b*Sech[c+d*x])^(n+1)/(b*d*(n+1)) +
  f*m/(b*d*(n+1))*Int[(e+f*x)^(m-1)*(a+b*Sech[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

```
Int[(e_.+f_.*x_)^m_.*Csch[c_.+d_.*x_]*Coth[c_.+d_.*x_]*(a_+b_.*Csch[c_.+d_.*x_])^n_.,x_Symbol] :=
  -(e+f*x)^m*(a+b*Csch[c+d*x])^(n+1)/(b*d*(n+1)) +
  f*m/(b*d*(n+1))*Int[(e+f*x)^(m-1)*(a+b*Csch[c+d*x])^(n+1),x] /;
FreeQ[{a,b,c,d,e,f,n},x] && IGtQ[m,0] && NeQ[n,-1]
```

5: $\int (e+f\,x)^m \operatorname{Sinh}[a+b\,x]^p \operatorname{Sinh}[c+d\,x]^q \,dx$ when $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+ \land m \in \mathbb{Z}$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+ \land m \in \mathbb{Z}$, then

$$\int (e+f\,x)^m \operatorname{Sinh}[a+b\,x]^p \operatorname{Cosh}[c+d\,x]^q \,dx \to \int (e+f\,x)^m \operatorname{TrigReduce}[\operatorname{Sinh}[a+b\,x]^p \operatorname{Cosh}[c+d\,x]^q] \,dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sinh[a_.+b_.*x_]^p_.*Sinh[c_.+d_.*x_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[(e+f*x)^m,Sinh[a+b*x]^p*Sinh[c+d*x]^q,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && IGtQ[q,0] && IntegerQ[m]
```

```
Int[(e_.+f_.*x_)^m_.*Cosh[a_.+b_.*x_]^p_.*Cosh[c_.+d_.*x_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[(e+f*x)^m,Cosh[a+b*x]^p*Cosh[c+d*x]^q,x],x] /;
FreeQ[{a,b,c,d,e,f},x] && IGtQ[p,0] && IGtQ[q,0] && IntegerQ[m]
```

6: $\int (e+f x)^m \operatorname{Sinh}[a+b x]^p \operatorname{Cosh}[c+d x]^q \, dx$ when $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+$, then

$$\int (e+f x)^m \operatorname{Sinh}[a+b x]^p \operatorname{Cosh}[c+d x]^q \, dx \to \int (e+f x)^m \operatorname{TrigReduce}\left[\operatorname{Sinh}[a+b x]^p \operatorname{Cosh}[c+d x]^q\right] dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*Sinh[a_.+b_.*x_]^p_.*Cosh[c_.+d_.*x_]^q_.,x_Symbol] :=
  Int[ExpandTrigReduce[(e+f*x)^m,Sinh[a+b*x]^p*Cosh[c+d*x]^q,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && IGtQ[p,0] && IGtQ[q,0]
```

7: $\int (e+f x)^m \operatorname{Sinh}[a+b x]^p \operatorname{Sech}[c+d x]^q \, dx$ when $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+ \land b c - a d == 0 \land \frac{b}{d} - 1 \in \mathbb{Z}^+$

Derivation: Algebraic expansion

Rule: If $p \in \mathbb{Z}^+ \land q \in \mathbb{Z}^+ \land b c - a d == 0 \land \frac{b}{d} - 1 \in \mathbb{Z}^+$, then

$$\int (e+f x)^m \operatorname{Sinh}[a+b x]^p \operatorname{Sech}[c+d x]^q \, dx \to \int (e+f x)^m \operatorname{TrigExpand}\left[\operatorname{Sinh}[a+b x]^p \operatorname{Cosh}[c+d x]^q\right] dx$$

Program code:

```
Int[(e_.+f_.*x_)^m_.*F_[a_.+b_.*x_]^p_.*G_[c_.+d_.*x_]^q_.,x_Symbol] :=
  Int[ExpandTrigExpand[(e+f*x)^m*G[c+d*x]^q,F,c+d*x,p,b/d,x],x] /;
FreeQ[{a,b,c,d,e,f,m},x] && MemberQ[{Sinh,Cosh},F] && MemberQ[{Sech,Csch},G] && IGtQ[p,0] && IGtQ[q,0] && EqQ[b*c-a*d,0] && IGtQ[b/d
```